

## Natural for Mainframes

ステートメント

バージョン 4.2.5

October 2009

This document applies to Natural バージョン 4.2.5 and to all subsequent releases.

Specifications contained herein are subject to change and these changes will be reported in subsequent release notes or new editions.

Copyright © Software AG 1979-2009. All rights reserved.

The name Software AG™, webMethods™, Adabas™, Natural™, ApplinX™, EntireX™ and/or all Software AG product names are either trademarks or registered trademarks of Software AG and/or Software AG USA, Inc. Other company and product names mentioned herein may be trademarks of their respective owners.

# 目次

1	ステートメント	1
2	構文記号およびオペランド定義テーブル	3
	構文記号	4
	オペランド定義テーブル	5
3	ステートメント使用関連トピック	9
4	機能別ステートメント	11
	データベースへのアクセスと更新	12
	算術演算とデータ移動操作	14
	ループ実行	14
	出力レポートの作成	15
	対話型処理用の画面生成	15
	論理条件の処理	16
	プログラムおよびルーチンの呼び出し	16
	プログラムとセッションの終了	17
	ワークファイル/PC ファイルの制御	17
	コンポーネントベースプログラミング	17
	ダイナミック変数または X-array のメモリ管理制御	18
	Natural リモートプロシージャコール	18
	インターネットと XML	18
	その他	18
	レポートモードのステートメント	19
	Predict Case および Entire DB で使用可能なステートメント	20
5	ACCEPT/REJECT	21
	機能	22
	構文説明	22
	複数の ACCEPT/REJECT ステートメントの処理	23
	リミット表記	23
	ホールド状態	23
	例	24
6	ADD	27
	機能	28
	構文説明	28
	例	30
7	ASSIGN	33
8	AT BREAK	35
	機能	36
	構文説明	37
	複数ブレイクレベルの階層	38
	例	39
9	AT END OF DATA	43
	機能	44
	制限事項	45
	構文説明	45

例 .....	46
10 AT END OF PAGE .....	49
機能 .....	50
構文説明 .....	52
例 .....	52
11 AT START OF DATA .....	55
機能 .....	56
構文説明 .....	57
例 .....	57
12 AT TOP OF PAGE .....	61
機能 .....	62
制限事項 .....	63
構文説明 .....	63
例 .....	63
13 BACKOUT TRANSACTION .....	65
機能 .....	66
制限事項 .....	67
データベース固有の考慮事項 .....	67
例 .....	67
14 BEFORE BREAK PROCESSING .....	69
機能 .....	70
制限事項 .....	71
構文説明 .....	71
例 .....	71
15 CALL .....	73
機能 .....	74
構文説明 .....	74
リターンコード .....	75
レジスタの使用方法 .....	75
境界の配置 .....	76
Adabas コール .....	77
ダイレクトロードとダイナミックロード .....	77
リンケージの規則 .....	79
PL/I プログラムのコール .....	81
C プログラムのコール .....	84
INTERFACE4 .....	87
16 CALL FILE .....	101
機能 .....	102
制限事項 .....	102
構文説明 .....	103
例 .....	103
17 CALL LOOP .....	105
機能 .....	106
制限事項 .....	106
構文説明 .....	107

例 .....	107
18 CALLNAT .....	109
機能 .....	110
構文説明 .....	111
ダイナミック変数を使用したパラメータ引き渡し .....	113
例 .....	114
19 CLOSE CONVERSATION .....	117
機能 .....	118
構文説明 .....	118
詳細と例 .....	119
20 CLOSE PC FILE .....	121
機能 .....	122
構文説明 .....	122
例 .....	122
21 CLOSE PRINTER .....	125
機能 .....	126
構文説明 .....	126
例 .....	127
22 CLOSE WORK FILE .....	129
機能 .....	130
構文説明 .....	130
例 .....	130
23 COMPOSE .....	133
機能 .....	134
構文説明 .....	135
フォーマット処理 .....	145
ダイアログモード .....	146
Natural 以外のプログラム .....	147
例 .....	149
24 COMPRESS .....	155
機能 .....	156
構文説明 .....	156
処理 .....	160
例 .....	160
25 COMPUTE .....	165
機能 .....	166
構文説明 .....	168
除算結果の精度 .....	170
SUBSTRING オプション .....	170
例 .....	170
26 CREATE OBJECT .....	173
機能 .....	174
構文説明 .....	174
27 DECIDE FOR .....	177
機能 .....	178

構文説明 .....	178
例 .....	179
28 DECIDE ON .....	181
機能 .....	182
構文説明 .....	182
例 .....	183
29 DEFINE CLASS .....	187
機能 .....	188
構文説明 .....	188
30 DEFINE DATA .....	191
31 構文の概要 .....	193
全般的な構文 .....	194
基本的な構文要素 .....	194
32 DEFINE DATA - 全般 .....	199
機能 .....	200
規則 .....	200
プログラミングモード .....	200
詳細な情報 .....	201
33 ローカルデータの定義 .....	203
機能 .....	204
制限事項 .....	204
構文説明 .....	204
34 グローバルデータの定義 .....	207
機能 .....	208
構文説明 .....	208
35 パラメータデータの定義 .....	209
機能 .....	210
制限事項 .....	210
構文説明 .....	210
36 アプリケーションに依存しない変数の定義 .....	215
機能 .....	216
構文説明 .....	216
37 Natural RPC 用のコンテキスト変数の定義 .....	219
機能 .....	220
制限事項 .....	221
構文説明 .....	221
38 NaturalX オブジェクトの定義 .....	223
機能 .....	224
構文説明 .....	224
39 変数定義 .....	227
機能 .....	228
構文説明 .....	228
40 ビューの定義 .....	231
機能 .....	232
構文説明 .....	232

41 再定義 .....	237
機能 .....	238
制限事項 .....	238
構文説明 .....	239
42 ハンドルの定義 .....	241
機能 .....	242
構文説明 .....	242
43 配列の次元の定義 .....	245
機能 .....	246
構文説明 .....	246
44 初期値の定義 .....	249
機能 .....	250
制限事項 .....	250
構文説明 .....	250
45 配列用の初期値/定数値 .....	253
機能 .....	254
制限事項 .....	254
構文説明 .....	255
46 フィールド/変数の EM、HD、PM パラメータ .....	257
機能 .....	258
構文説明 .....	258
47 DEFINE DATA ステートメントの使用例 .....	259
例 1 - DEFINE DATA LOCAL (ダイレクトデータ定義) .....	260
例 2 - DEFINE DATA LOCAL (配列の定義/初期化) .....	260
例 3 - DEFINE DATA (ビューの定義、配列の再定義) .....	261
例 4 - DEFINE DATA (グローバル、パラメータ、およびローカルデータエリ ア) .....	262
例 5 - DEFINE DATA (初期化) .....	263
例 6 - DEFINE DATA (可変配列) .....	264
48 DEFINE PRINTER .....	267
機能 .....	268
構文説明 .....	269
z/OS バッチ、TSO、およびサーバーでのプリンタ名 .....	272
z/VSE バッチ環境でのプリンタ名 .....	275
VM/CMS 環境でのプリンタ名 .....	276
BS2000/OSD バッチおよび TIAM 環境でのプリンタ名 .....	277
CICS 環境でのプリンタ名 .....	283
Com-plete 環境でのプリンタ名 .....	283
Com-plete/SMARTS 環境でのプリンタ名 .....	283
Natural Advanced Facilities 環境でのプリンタ名 .....	284
追加レポートのプリンタ名 .....	284
例 .....	285
49 DEFINE SUBROUTINE .....	287
機能 .....	288
制限事項 .....	288

構文説明 .....	289
サブルーチンで使用可能なデータ .....	290
例 .....	290
50 DEFINE WINDOW .....	293
機能 .....	294
構文説明 .....	295
ウィンドウの入力フィールドの保護 .....	299
他のウィンドウの呼び出し .....	299
例 .....	299
51 DEFINE WORK FILE .....	301
機能 .....	302
構文説明 .....	302
z/OS バッチ、TSO、およびサーバーでのワークファイル名 .....	304
z/VSE バッチでのワークファイル名 .....	307
VM/CMS でのワークファイル名 .....	308
BS2000/OSD バッチおよび TIAM でのワークファイル名 .....	309
CICS でのワークファイル名 .....	313
Com-plete/SMARTS でのワークファイル名 .....	313
52 DELETE .....	315
機能 .....	316
制限事項 .....	316
構文説明 .....	316
データベース固有の考慮事項 .....	317
例 .....	317
53 DISPLAY .....	319
機能 .....	320
構文説明 .....	320
DISPLAY ステートメントに適用されるデフォルト .....	332
例 .....	333
54 DIVIDE .....	341
機能 .....	342
構文説明 .....	342
例 .....	345
55 DO/DOEND .....	347
機能 .....	348
制限事項 .....	348
例 .....	349
56 DOWNLOAD PC FILE .....	351
機能 .....	352
構文説明 .....	352
例 .....	353
57 EJECT .....	355
機能 .....	356
構文説明 .....	356
処理 .....	358



例 .....	358
58 END .....	361
機能 .....	362
構文説明 .....	362
例 .....	363
59 END TRANSACTION .....	365
機能 .....	366
制限事項 .....	366
構文説明 .....	367
関連データベース .....	367
データベース固有の考慮事項 .....	368
例 .....	368
60 ESCAPE .....	371
機能 .....	372
構文説明 .....	373
例 .....	374
61 EXAMINE .....	377
構文 1 - EXAMINE .....	378
構文 2 - EXAMINE TRANSLATE .....	383
構文 3 - Unicode 書記素用の EXAMINE .....	384
例 .....	386
62 EXPAND .....	393
機能 .....	394
構文説明 .....	394
63 FETCH .....	399
機能 .....	400
構文説明 .....	401
例 .....	402
64 FIND .....	405
機能 .....	406
制限 .....	408
構文説明 .....	408
例 .....	430
65 FOR .....	441
機能 .....	442
構文説明 .....	442
例 .....	443
66 FORMAT .....	445
機能 .....	446
構文説明 .....	446
適用可能なパラメータ .....	447
例 .....	448
67 GET .....	451
機能 .....	452
制限 .....	452

構文説明 .....	453
例 .....	454
68 GET SAME .....	457
機能 .....	458
制限 .....	458
構文説明 .....	459
例 .....	459
69 GET TRANSACTION DATA .....	461
機能 .....	462
制限事項 .....	463
構文説明 .....	463
例 .....	463
70 HISTOGRAM .....	465
機能 .....	466
制限 .....	467
構文説明 .....	467
例 .....	473
71 IF .....	477
機能 .....	478
構文説明 .....	478
例 .....	479
72 IF SELECTION .....	481
機能 .....	482
構文説明 .....	482
例 .....	483
73 IGNORE .....	485
機能 .....	486
例 .....	486
74 INCLUDE .....	487
機能 .....	488
構文説明 .....	488
例 .....	489
75 INPUT .....	495
機能 .....	496
入力モード .....	496
データ入力 .....	498
SB - 選択ボックス .....	500
エラー修正 .....	500
画面分割 .....	501
76 INPUT 構文 1 - ダイナミック画面レイアウトの指定 .....	503
INPUT 構文 1 - 説明 .....	504
例 - 構文 1 .....	513
77 INPUT 構文 2 - 定義済みマップレイアウトの使用 .....	517
パラメータリストのない INPUT USING MAP .....	518
プログラムで定義されている INPUT フィールド .....	519

INPUT 構文 2 - 説明 .....	519
非スクリーンモードでの INPUT ステートメントの使用 .....	520
Natural スタックデータの処理 .....	522
バッチモードでの INPUT ステートメントの使用 .....	522
78 INTERFACE .....	525
機能 .....	526
構文説明 .....	527
79 LIMIT .....	533
機能 .....	534
構文説明 .....	535
例 .....	535
80 LOOP .....	537
機能 .....	538
制限事項 .....	538
構文説明 .....	538
例 .....	539
81 METHOD .....	541
機能 .....	542
構文説明 .....	542
例 .....	543
82 MOVE .....	547
機能 .....	548
構文説明 .....	549
例 .....	560
83 MOVE ALL .....	565
機能 .....	566
構文説明 .....	566
例 .....	567
84 MOVE INDEXED .....	569
85 MULTIPLY .....	571
機能 .....	572
構文説明 .....	572
例 .....	574
86 NEWPAGE .....	577
機能 .....	578
構文説明 .....	578
例 .....	579
87 OBTAIN .....	583
機能 .....	584
制限事項 .....	584
構文説明 .....	585
例 .....	589
88 ON ERROR .....	591
機能 .....	592
制限事項 .....	592

構文説明 .....	593
サブルーチン内の ON ERROR 処理 .....	593
システム変数 *ERROR-NR および *ERROR-LINE .....	593
例 .....	594
89 OPEN CONVERSATION .....	595
機能 .....	596
構文説明 .....	596
詳細と例 .....	597
90 OPTIONS .....	599
機能 .....	600
複数の OPTIONS ステートメントの処理 .....	600
91 PARSE XML .....	601
機能 .....	602
構文説明 .....	603
例 .....	605
92 PASSW .....	611
機能 .....	612
制限事項 .....	613
構文説明 .....	613
例 .....	614
93 PERFORM .....	615
機能 .....	616
構文説明 .....	617
例 .....	619
94 PERFORM BREAK PROCESSING .....	623
機能 .....	624
構文説明 .....	624
例 .....	625
95 PRINT .....	627
機能 .....	628
構文説明 .....	629
例 .....	634
96 PROCESS .....	637
機能 .....	638
制限事項 .....	638
構文説明 .....	638
97 PROCESS COMMAND .....	641
機能 .....	643
構文説明 .....	643
DDM : COMMAND .....	654
例 .....	655
98 PROCESS PAGE .....	657
機能 .....	658
構文 1 - PROCESS PAGE .....	658
構文 2 - PROCESS PAGE USING .....	660

構文 3 - PROCESS PAGE UPDATE .....	662
構文 4 - PROCESS PAGE MODAL .....	665
99 PROPERTY .....	669
機能 .....	670
構文説明 .....	670
例 .....	671
100 READ .....	673
機能 .....	674
構文説明 .....	675
READ で使用可能なシステム変数 .....	684
例 .....	685
101 READ WORK FILE .....	693
機能 .....	694
構文説明 .....	695
フィールド長 .....	698
ラージおよびダイナミック変数の処理 .....	698
例 .....	699
102 REDEFINE .....	701
機能 .....	702
制限事項 .....	702
構文説明 .....	702
例 .....	703
103 REDUCE .....	705
機能 .....	706
構文説明 .....	706
104 REINPUT .....	711
機能 .....	712
構文説明 .....	713
例 .....	719
105 REJECT .....	723
106 RELEASE .....	725
機能 .....	726
構文説明 .....	726
例 .....	727
107 REPEAT .....	729
機能 .....	730
構文説明 .....	730
例 .....	731
108 REQUEST DOCUMENT .....	735
機能 .....	736
構文説明 .....	737
受信／送信データのエンコード .....	745
例 .....	747
109 RESET .....	751
機能 .....	752

構文説明 .....	752
例 .....	753
110 RESIZE .....	755
機能 .....	756
構文説明 .....	756
111 RETRY .....	761
機能 .....	762
制限事項 .....	762
例 .....	762
112 RUN .....	765
機能 .....	766
構文説明 .....	766
ダイナミックなソーステキストの作成／実行 .....	767
例 .....	768
113 SEND METHOD .....	771
機能 .....	772
構文説明 .....	772
例 .....	775
114 SEPARATE .....	781
機能 .....	782
構文説明 .....	782
例 .....	785
115 SET CONTROL .....	789
機能 .....	790
構文説明 .....	790
例 .....	790
116 SET GLOBALS .....	793
機能 .....	794
パラメータ .....	794
例 .....	796
117 SET KEY .....	797
機能 .....	798
構文説明 .....	798
キーのプログラム察知の有効化／キーの解除 .....	800
コマンド／プログラムの割り当て .....	801
入力データの割り当て .....	802
COMMAND OFF/ON .....	802
HELP の割り当て .....	803
DYNAMIC オプション .....	803
DISABLED オプション .....	804
異なるプログラムレベルでの SET KEY ステートメント .....	804
名前の割り当て .....	805
例 .....	807
118 SET TIME .....	809
機能 .....	810

例 .....	810
119 SET WINDOW .....	813
機能 .....	814
構文説明 .....	814
例 .....	814
120 SKIP .....	815
機能 .....	816
構文説明 .....	816
例 .....	817
121 SORT .....	819
機能 .....	820
制限事項 .....	821
構文説明 .....	821
SORT ステートメント処理 (3 段階) .....	823
例 .....	824
外部ソートプログラムの使用 .....	828
122 STACK .....	829
機能 .....	830
構文説明 .....	830
例 .....	832
123 STOP .....	835
機能 .....	836
例 .....	836
124 STORE .....	839
機能 .....	840
データベース固有の考慮事項 .....	841
構文説明 .....	841
例 .....	843
125 SUBTRACT .....	847
機能 .....	848
構文説明 .....	848
例 .....	850
126 SUSPEND IDENTICAL SUPPRESS .....	851
機能 .....	852
構文説明 .....	852
例 .....	852
127 TERMINATE .....	857
機能 .....	858
構文説明 .....	858
Natural 終了後に制御を受け取るプログラム .....	859
例 .....	859
128 UPDATE .....	861
機能 .....	862
制限事項 .....	863
データベース固有の考慮事項 .....	863

構文説明 .....	864
例 .....	865
129 UPLOAD PC FILE .....	867
機能 .....	869
構文説明 .....	869
例 .....	870
130 WRITE .....	871
機能 .....	872
構文1 - 直接画面定義 .....	872
構文 1 - 説明 .....	873
構文 2 - フォーム／マップ使用 .....	881
構文 2 - 説明 .....	881
例 .....	882
131 WRITE TITLE .....	887
機能 .....	888
制限事項 .....	889
構文説明 .....	889
例 .....	892
132 WRITE TRAILER .....	895
機能 .....	896
制約 .....	897
構文説明 .....	897
例 .....	900
133 WRITE WORK FILE .....	903
機能 .....	904
構文説明 .....	904
フィールドの外部表示 .....	905
ラージおよびダイナミック変数の処理 .....	906
例 .....	907
134 SQL ステートメント .....	909
135 一般セットと拡張セット .....	911
136 基本構文項目 .....	913
定数 .....	914
名前 .....	914
パラメータ .....	917
Natural フォーマットと SQL データタイプ .....	920
137 Natural ビューの概念 .....	923
138 スカラー式 .....	925
スカラー式 .....	926
スカラー演算子 .....	926
ファクタ .....	927
139 検索条件 .....	933
検索条件 .....	934
属性 .....	934
140 選択式 .....	939









選択 .....	940
テーブル式 .....	941
141 フレキシブル SQL .....	945
フレキシブル SQL の使用 .....	946
フレキシブル SQL でのテキスト変数の指定 .....	947
142 CALLDBPROC - SQL .....	951
機能 .....	952
制限事項 .....	953
構文説明 .....	953
例 .....	955
143 COMMIT - SQL .....	957
機能 .....	958
Natural 以外のプログラムについて .....	958
例 .....	958
144 DELETE - SQL .....	959
機能 .....	960
構文説明 .....	960
145 INSERT - SQL .....	963
機能 .....	964
構文説明 .....	964
例 .....	969
146 PROCESS SQL .....	971
機能 .....	972
構文説明 .....	972
例 .....	973
147 READ RESULT SET - SQL .....	975
機能 .....	976
制限事項 .....	976
構文説明 .....	977
例 .....	978
148 ROLLBACK - SQL .....	979
機能 .....	980
Natural 以外のプログラムについて .....	980
例 .....	981
149 SELECT - SQL .....	983
機能 .....	984
構文説明 .....	984
ジョインクエリ .....	997
SELECT - カーソル指向 .....	997
150 UPDATE - SQL .....	1007
機能 .....	1008
構文説明 .....	1008
例 .....	1011
151 参照プログラム例 .....	1013
ASSIGN .....	1014

AT BREAK .....	1015
AT END OF DATA .....	1017
AT END OF PAGE .....	1018
AT START OF DATA .....	1018
AT TOP OF PAGE .....	1020
DEFINE SUBROUTINE .....	1021
FIND .....	1022
FOR .....	1024
HISTOGRAM .....	1025
IF .....	1025
PERFORM BREAK PROCESSING .....	1027
READ .....	1028
REPEAT .....	1029
SORT .....	1030
STORE .....	1031
UPDATE .....	1033
システム変数のプログラム例 .....	1034
索引 .....	1039

# 1 ステートメント

このドキュメントでは、Natural プログラミング言語のステートメントについて説明します。次の項目で構成されています。

 <b>構文記号およびオペランド定義テーブル</b>	ステートメントの構文を示す図で使用している記号、およびオペランド定義テーブルに関する情報を示します。
 <b>ステートメント使用関連トピック</b>	『プログラミングガイド』の特別なトピックのリストを示します。
 <b>機能別ステートメント</b>	機能別に分類された Natural ステートメントの概要を示します。
 <b>ステートメント (アルファベット順)</b>	ステートメント (SQL ステートメントを除く) をアルファベット順で説明します。
 <b>Natural SQL ステートメント</b>	SQL データベース内のデータを管理するために Natural プログラムで使用できる特定のステートメントについて説明します。
 <b>参照プログラム例</b>	Natural ステートメントおよびシステム変数のリファレンスドキュメントで参照されている追加のプログラム例を示します。  <b>注意:</b> 一般に、ステートメントの説明で示されているプログラム例はストラクチャードモードで記述されています。レポーティングモードとストラクチャードモードで構文が大幅に異なるステートメントについては、同等のレポーティングモードの例も参照として提供されています。これらのプログラム例は、Natural ライブラリ SYSESYN にソースコード形式で用意されています。Natural ステートメントのその他のプログラム例については、『プログラミングガイド』の「参照プログラム例」を参照してください。これらのプログラム例は、Natural ライブラリ SYSEXPB に用意されています。これらのライブラリが使用可能であるかを Natural 管理者に確認してください。プログラム例では、Software AG 提供のデモ用ファイル EMPLOYEES と VEHICLES のデータを使用します。



## 2 構文記号およびオペランド定義テーブル

---

▪ 構文記号 .....	4
▪ オペランド定義テーブル .....	5

## 構文記号

Natural ステートメントの構文を示す図には次の記号を使用しています。

構文記号	説明
ABCDEF	大文字の用語は Natural キーワードまたは Natural 予約語のいずれかで、示されたとおりに入力する必要があります。
<u>ABCDEF</u>	大文字のオプション用語全体に下線が付いている（ハイパーリンクではない）場合は、その用語がデフォルト値であることを示します。用語を省略すると、下線の付いた値が適用されます。
<u>ABCDEF</u>	大文字の用語の一部に下線が付いている（ハイパーリンクではない）場合は、下線部分はその用語の入力可能な省略形です。
<i>abcdef</i>	斜体の文字は、変数情報を表します。この用語を指定する場合は、有効な値を入力する必要があります。 <b>注意:</b> <i>statement</i> または <i>statements</i> の代わりに、状況に応じて 1 つまたは複数のステートメントを指定する必要があります。不要なステートメントがある場合は、 <b>IGNORE</b> ステートメントを挿入します。
[ ]	角カッコ内の要素はオプションです。  角カッコに複数の行が入れ子になって含まれている場合は、各行がオプションの選択肢です。選択肢の中から 1 つ選択できます。
{ }	中カッコに複数の行が入れ子になって含まれている場合は、各行が選択肢です。選択肢の中から 1 つのみ選択する必要があります。
	選択肢は縦棒で区切られます。
...	省略記号の前の用語は任意に繰り返すことができます。省略記号の後の数値は用語の繰り返し回数を示します。  省略記号の前の用語が角カッコまたは中カッコに囲まれている場合、省略記号はカッコ全体に適用します。
,...	コンマと省略記号の前の用語は任意に繰り返すことができます。繰り返す場合は、繰り返しをコンマで区切る必要があります。コンマと省略記号の後の数値は用語の繰り返し回数を示します。  コンマと省略記号の前の用語が角カッコまたは中カッコに囲まれている場合、コンマと省略記号はカッコ全体に適用します。
:...	コロンと省略記号の前の用語は任意に繰り返すことができます。繰り返す場合は、繰り返しをコロンの区切る必要があります。コロンと省略記号の後の数値は用語の繰り返し回数を示します。  コロンと省略記号の前の用語が角カッコまたは中カッコに囲まれている場合、コロンと省略記号はカッコ全体に適用します。

構文記号	説明
他の記号  ( [ ] { }   ... .. ... を除く)	他のすべての記号は、この表で定義しているものを除いて、示されたとおりに入力する必要があります。  例外：SQL スカラー連結演算子は 2 つの縦線で表現され、構文定義に示されたとおりに完全に入力する必要があります。

例：

```
WRITE [USING] { FORM }
                { MAP } operand1 [operand2 ...]
```

- WRITE、USING、MAP、および FORM は Natural キーワードであり、示されたとおりに入力する必要があります。
- operand1 および operand2 はユーザーが指定する変数であり、処理するオブジェクトの名前を指定します。
- 中カッコは、FORM または MAP のうちどちらか 1 つを必ず選択する必要があることを示しています。
- 角カッコは、USING および operand2 がオプションの要素であり、必要に応じて指定できることを示しています。
- 省略記号は、operand2 を複数回指定できることを示しています。

## オペランド定義テーブル

Natural ステートメントの構文に 1 つ以上のオペランドが含まれる場合、次の形式の表が示されています。

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
operand1	C S A G N/M E A U N P I F B D T L C O		可/不可	可/不可

表内の意味は次のとおりです。

## 構文要素

オペランドで使用できる要素を示します。

<b>C</b>	定数
<b>S</b>	単一オカレンス（スカラ、つまり配列でもグループでもないフィールド／変数）
<b>A</b>	配列
<b>G</b>	グループ
<b>N/M</b>	Natural システム変数 N = 全システム変数を使用できることを示します。 M = 変更可能なシステム変数のみを使用できることを示します。システム変数の内容が変更可能かどうかについては、Natural の『システム変数』ドキュメントを参照してください。
<b>E</b>	算術演算式

## フォーマット

オペランドで使用できるフォーマットを示します。

<b>A</b>	英数字（EBCDIC コードページ）
<b>U</b>	英数字（Unicode）
<b>N</b>	アンパック型数値
<b>P</b>	パック型数値
<b>I</b>	整数
<b>F</b>	浮動小数点
<b>B</b>	バイナリ
<b>D</b>	日付
<b>T</b>	時刻
<b>L</b>	論理
<b>C</b>	属性制御
<b>O</b>	HANDLE OF OBJECT



## ステートメント参照

ステートメントラベルまたはソースコード行番号で参照できるかどうかを示します。

## ダイナミック定義

プログラム内でフィールドをダイナミックに定義できるかどうかを示します。レポーティングモードでのみ可能です。



# 3 ステートメント使用関連トピック

---

Natural ステートメントの使用方法の詳細については、Natural の『プログラミングガイド』で次の情報を参照してください。

- ユーザー定義変数
- ダイナミック変数／フィールドとラージ変数／フィールド
  - はじめに
  - 使用方法
- ユーザー定義定数
- レポート指定
- テキスト表記
- ユーザーコメント
- 論理条件基準
- 演算割り当てのルール




## 4 機能別ステートメント

---

■ データベースへのアクセスと更新 .....	12
■ 算術演算とデータ移動操作 .....	14
■ ループ実行 .....	14
■ 出力レポートの作成 .....	15
■ 対話型処理用の画面生成 .....	15
■ 論理条件の処理 .....	16
■ プログラムおよびルーチンの呼び出し .....	16
■ プログラムとセッションの終了 .....	17
■ ワークファイル/PC ファイルの制御 .....	17
■ コンポーネントベースプログラミング .....	17
■ ダイナミック変数または X-array のメモリ管理制御 .....	18
■ Natural リモートプロシージャコール .....	18
■ インターネットと XML .....	18
■ その他 .....	18
■ レポートモードのステートメント .....	19
■ Predict Case および Entire DB で使用可能なステートメント .....	20

この章では、機能別に分類されたステートメントの概要を示します。

 **注意:**

1. ストラクチャードモードおよびレポーティングモードの両方で使用できるステートメントと、レポーティングモードでしか使用できないステートメントが存在します。『プログラミングガイド』の「*Natural* プログラミングモード」を参照してください。
2. DLOGOFF、DLOGON、SHOW、IMPORT、およびEXPORT ステートメントは、Entire DBがインストールされている場合にのみ使用できます。詳細については、『Entire DB』ドキュメントを参照してください。

## データベースへのアクセスと更新

---

### Natural DML ステートメント

データベース内の情報にアクセスして操作するには、次の Natural データ操作言語 (DML) ステートメントを使用します。

<b>READ</b>	データベースファイル内の物理的または論理的な一連のレコードを読み取ります。
<b>FIND</b>	ユーザー指定の条件に基づいて、データベースファイルからレコードを選択します。
<b>HISTOGRAM</b>	データベースフィールドの値を読み取ります。
<b>GET</b>	ISN (内部シーケンス番号) または RNO (レコード番号) を指定してレコードを読み取ります。
<b>GET SAME</b>	現在処理中のレコードを再読み取りします。
<b>ACCEPT/REJECT</b>	ユーザー指定の条件に基づいて、レコードの受け入れ/除外を行います。
<b>PASSW</b>	パスワード保護されたファイルにアクセスするためのパスワードを指定します。
<b>LIMIT</b>	READ、FIND、または HISTOGRAM 処理ループの実行回数を制限します。
<b>STORE</b>	新しいレコードをデータベースに追加します。
<b>UPDATE</b>	データベース内のレコードを更新します。
<b>DELETE</b>	データベースからレコードを削除します。
<b>END TRANSACTION</b>	論理トランザクションの終了を示します。
<b>BACKOUT TRANSACTION</b>	完了していない論理トランザクションをバックアウトします。
<b>GET TRANSACTION DATA</b>	直前の END TRANSACTION ステートメントで保存したトランザクションデータを読み取ります。

<b>RETRY</b>	他のユーザー用にホールド状態になっているレコードの再読み取りを試みます。
<b>AT START OF DATA</b>	処理ループで一連のレコードの先頭が処理されるときに実行されるステートメントを指定します。
<b>AT END OF DATA</b>	処理ループで一連のレコードの最後が処理されたときに実行されるステートメントを指定します。
<b>AT BREAK</b>	制御フィールドの値が変更されるときに実行されるステートメントを指定します（ブレイク処理）。
<b>BEFORE BREAK PROCESSING</b>	ブレイク処理の実行前に実行されるステートメントを指定します。
<b>PERFORM BREAK PROCESSING</b>	直ちにブレイク処理を呼び出します。

### Natural SQL ステートメント

Natural DML ステートメントの他に、Natural プログラムで SQL ステートメントを使用して直接 SQL を扱うことができます。

次の SQL ステートメントが有効です。

<b>CALLDBPROC</b>	Natural からアクセスできる SQL データベースシステムのストアードプロシージャを呼び出します。
<b>COMMIT</b>	論理トランザクションの終了を示し、トランザクション中にロックされた全データを解放します。すべてのデータ変更がコミットされて確定されます。
<b>DELETE</b>	カーソルを使用しないでテーブル内の行を削除したり（「検索済」 DELETE）、カーソルが位置づけられたテーブル内の行を削除したり（「位置決め」 DELETE）します。
<b>INSERT</b>	1 つ以上の行をテーブルに新しく追加します。
<b>PROCESS SQL</b>	基準データベースに対して SQL ステートメントを発行します。
<b>READ RESULT SET</b>	前の CALLDBPROC ステートメントで呼び出されたストアードプロシージャによって作成された結果セットを読み込みます。
<b>ROLLBACK</b>	最後のリカバリ単位の開始以降に行われたデータベース更新をすべて元に戻します。
<b>SELECT</b>	任意の行数を取得するためのカーソル選択と、多くても 1 つの行を取得する非カーソル選択（単独 SELECT）の両方をサポートします。
<b>UPDATE</b>	カーソルを使用しないでテーブル内の行に対して更新処理を実行したり（「検索済」 UPDATE）、カーソルが位置づけられた行の列に対して更新処理を実行したり（「位置決め」 UPDATE）します。

## 算術演算とデータ移動操作

算術演算とデータ移動操作には次のステートメントを使用します。

<b>COMPUTE</b>	算術演算を実行するか、値をフィールドに割り当てます。
<b>ADD</b>	複数のオペランドを加算します。
<b>SUBTRACT</b>	1 オペランドから他の1つ以上のオペランドを減算します。
<b>MULTIPLY</b>	複数のオペランドを乗算します。
<b>DIVIDE</b>	1 オペランドを他のオペランドに除算します。
<b>EXAMINE TRANSLATE</b>	フィールド内の文字を大文字/小文字または他の文字に変換します。
<b>MOVE</b>	1 オペランドの値を1つ以上のフィールドに移動します。
<b>MOVE ALL</b>	値の複数オカレンスを他のフィールドに移動します。
<b>COMPRESS</b>	複数のフィールドの値を1フィールドに連結します。
<b>SEPARATE</b>	1フィールドの内容を複数のフィールドに分割します。
<b>EXAMINE</b>	特定の値でフィールドをスキャンし、それを置き換えたり、出現回数をカウントしたりします。
<b>RESET</b>	フィールドの値を0（数値の場合）、空白（英数字の場合）、またはその初期値に設定します。

## ループ実行

処理ループの実行には次のステートメントが関係します。

<b>ESCAPE</b>	処理ループの実行を停止します。
<b>FOR</b>	処理ループを開始し、ループの処理回数を制御します。
<b>REPEAT</b>	処理ループを開始し、指定された条件に基づいて終了します。
<b>SORT</b>	レコードをソートします。



## 出力レポートの作成

出力レポートの作成には次のステートメントを使用します。

<b>FORMAT</b>	出力パラメータ設定を指定します。
<b>DISPLAY</b>	列形式で出力するフィールドを指定します。
<b>WRITE / PRINT</b>	列形式以外で出力するフィールドを指定します。
<b>WRITE TITLE</b>	レポートの各ページの上に出力するテキストを指定します。
<b>WRITE TRAILER</b>	レポートの各ページの下に出力するテキストを指定します。
<b>AT TOP OF PAGE</b>	新しい出力ページの開始時に実行される処理を指定します。
<b>AT END OF PAGE</b>	出力ページの最後に達したときに実行される処理を指定します。
<b>SKIP</b>	レポートに1行以上の空行を作成します。
<b>EJECT</b>	タイトルやヘッダーを含まずにページ送りを行います。
<b>NEWPAGE</b>	タイトルやヘッダー付きでページ送りを行います。
<b>SUSPEND IDENTICAL SUPPRESS</b>	1レコードに対して同一値の省略を中止します。
<b>DEFINE PRINTER</b>	論理出力先にレポートを割り当てます。
<b>CLOSE PRINTER</b>	プリンタをクローズします。

## 対話型処理用の画面生成

データの対話型処理のためにデータ画面（マップ）を生成するには、次のステートメントを使用します。

<b>INPUT</b>	データを表示したり、入力したりするための形式化された画面（マップ）を作成します。
<b>REINPUT</b>	INPUT ステートメントを再実行します（前の INPUT ステートメントに対する入力データが誤りであった場合）。
<b>DEFINE WINDOW</b>	ウィンドウのサイズ、位置、および属性を指定します。
<b>SET WINDOW</b>	ウィンドウを稼動したり解除したりします。
<b>PROCESS PAGE</b>	Web リッチ GUI 画面に対するデータマッピングを作成します。
<b>PROCESS PAGE USING</b>	ページレイアウトから生成されたアダプタオブジェクトを使用してリッチ GUI I/O 処理を実行します。
<b>PROCESS PAGE UPDATE</b>	PROCESS PAGE ステートメントを再実行します。
<b>PROCESS PAGE MODAL</b>	処理ブロックを開始し、リッチ GUI ウィンドウの存続期間を制御します。

## 論理条件の処理

Natural プログラムの実行中に検出された条件に基づいてステートメントの実行を制御するには、次のステートメントを使用します。

<b>IF</b>	論理条件に基づいてステートメントを実行します。
<b>IF SELECTION</b>	一連の英数字フィールドのうちの1つにのみ値が含まれていることを確認します。
<b>DECIDE FOR</b>	複数の論理条件に基づいてステートメントを実行します。
<b>DECIDE ON</b>	変数の内容に基づいてステートメントを実行します。
<b>ON ERROR</b>	ランタイムエラーをインターセプトします。インターセプトしない場合は、Natural エラーメッセージが表示された後、Natural プログラムが終了します。

## プログラムおよびルーチンの呼び出し

プログラムおよびルーチンの実行と関連して次のステートメントを使用します。

<b>CALL</b>	Natural プログラムから Natural 以外のプログラムを呼び出します。
<b>CALLNAT</b>	Natural サブプログラムを呼び出します。
<b>CALL FILE</b>	Natural 以外のプログラムを呼び出し、Adabas 以外のファイルからレコードを読み取ります。
<b>CALL LOOP</b>	Natural 以外のプログラムへのコールを含む処理ループを生成します。
<b>DEFINE SUBROUTINE</b>	Natural サブルーチンを定義します。
<b>ESCAPE</b>	ルーチンの実行を中止します。
<b>FETCH</b>	Natural プログラムを呼び出します。
<b>PERFORM</b>	Natural サブルーチンを呼び出します。
<b>PROCESS COMMAND</b>	コマンドプロセッサを呼び出します。
<b>RUN</b>	ソースプログラムをコンパイルして実行します。

## プログラムとセッションの終了

アプリケーションの実行または Natural セッションを終了するには、次の Natural ステートメントを使用します。

<b>STOP</b>	アプリケーションの実行を終了します。
<b>TERMINATE</b>	Natural セッションを終了します。

## ワークファイル／PC ファイルの制御

(Adabas以外の) 物理順ワークファイルに対するデータの読み取り／書き込みには次の Natural ステートメントを使用します。

<b>WRITE WORK FILE</b>	ワークファイルにデータを書き込みます。
<b>DOWNLOAD PC FILE</b>	メインフレーム、UNIX、または OpenVMS プラットフォームから PC へのデータ転送を可能にします。
<b>READ WORK FILE</b>	ワークファイルからデータを読み取ります。
<b>UPLOAD PC FILE</b>	PC からメインフレーム、UNIX、または OpenVMS プラットフォームへのデータ転送を可能にします。
<b>CLOSE WORK FILE</b>	ワークファイルをクローズします。
<b>CLOSE PC FILE</b>	特定の PC ワークファイルをクローズします。
<b>DEFINE WORK FILE</b>	ファイル名をワークファイルに割り当てます。

## コンポーネントベースプログラミング

コンポーネントベースプログラミングと関連して次の Natural ステートメントを使用します。

<b>DEFINE CLASS</b>	Natural クラスモジュール内からクラスを指定します。
<b>CREATE OBJECT</b>	指定したクラスのオブジェクト（インスタンスとも呼ばれる）を作成します。
<b>SEND METHOD</b>	オブジェクトのメソッドを呼び出します。
<b>INTERFACE</b>	クラスの特長機能に対するインターフェイス（メソッドおよびプロパティの集合）を定義します。
<b>METHOD</b>	インターフェイス定義外のメソッドの実装として、サブプログラムを割り当てます。
<b>PROPERTY</b>	インターフェイス定義外のプロパティに対する実装として、オブジェクトデータ変数を割り当てます。

## ダイナミック変数または X-array のメモリ管理制御

<b>EXPAND</b>	ダイナミック変数の割り当てメモリを指定したサイズに拡張します。
<b>REDUCE</b>	ダイナミック変数のサイズを縮小します。
<b>RESIZE</b>	ダイナミック変数のサイズを調整します。

## Natural リモートプロシージャコール

<b>OPEN CONVERSATION</b>	これにより、クライアントは会話を開き、会話に取り込むリモートサブプログラムを指定することができます。
<b>CLOSE CONVERSATION</b>	これにより、クライアントは会話を閉じることができます。現在の会話、開かれた別の会話、または開かれたすべての会話を閉じることができます。
<b>DEFINE DATA CONTEXT</b>	コンテキスト変数と呼ばれる変数を定義します。コンテキスト変数では、対応する CALLNAT ステートメントで変数をパラメータとして明示的に渡さなくても、1つの会話の中で複数のリモートサブプログラムがこの変数を使用できます。

『Natural リモートプロシージャコール』ドキュメントの「関係する Natural ステートメント」も参照してください。

## インターネットと XML

<b>PARSE</b>	Natural プログラムから XML ドキュメントを解析できるようになります。
<b>REQUEST DOCUMENT</b>	外部システムへのアクセスを可能にします。

## その他

<b>DEFINE DATA</b>	Natural プログラムまたはルーチンで使用するデータ要素を定義します。
<b>END</b>	Natural プログラムまたはルーチンのソースコードの終了を示します。
<b>INCLUDE</b>	コンパイル時に Natural コピーコードを組み込みます。
<b>RELEASE</b>	Natural スタックの内容を削除します。FIND ステートメントによって保存された ISN の集合を解放します。Natural グローバル変数を解放します。
<b>SET CONTROL</b>	Natural プログラム内から Natural 端末コマンドを実行します。
<b>SET KEY</b>	端末キーに機能を割り当てます。

<b>SET GLOBALS</b>	セッションパラメータの値を設定します。
<b>SET TIME</b>	*TIMD システム変数に対する参照開始時点を設定します。
<b>STACK</b>	Natural スタックにデータ/コマンドを挿入します。

## レポーティングモードのステートメント

次のステートメントは、レポーティングモードでのみ使用できます。

<b>CLOSE LOOP</b>	処理ループを閉じます。
<b>DO/DOEND</b>	論理条件に基づいて実行するステートメントのグループを指定します。
<b>OBTAIN</b>	ファイルから1つ以上のフィールドを読み取ります。
<b>REDEFINE</b>	フィールドを再定義します。

次のステートメントは、ストラクチャードモードとレポーティングモードの両方で使用できます。ただし、ステートメント構造と機能が異なるものがあります。

<b>AT START OF DATA</b>	処理ループで一連のレコードの先頭が処理されるときに実行されるステートメントを指定します。
<b>AT END OF DATA</b>	処理ループで一連のレコードの最後が処理されたときに実行されるステートメントを指定します。
<b>AT BREAK</b>	制御フィールドの値が変更されるときに実行されるステートメントを指定します（ブレイク処理）。
<b>AT TOP OF PAGE</b>	新しい出力ページの開始時に実行される処理を指定します。
<b>AT END OF PAGE</b>	出力ページの最後に達したときに実行される処理を指定します。
<b>BEFORE BREAK PROCESSING</b>	ブレイク処理の実行前に実行されるステートメントを指定します。
<b>CALL LOOP</b>	Natural 以外のプログラムへのコールを含む処理ループを生成します。
<b>CALL FILE</b>	Natural 以外のプログラムを呼び出し、Adabas 以外のファイルからレコードを読み取ります。
<b>COMPUTE</b>	算術演算を実行するか、値をフィールドに割り当てます。
<b>DEFINE SUBROUTINE</b>	Natural サブルーチンを定義します。
<b>ESCAPE</b>	処理ループの実行を停止します。
<b>FIND</b>	ユーザー指定の条件に基づいて、データベースファイルからレコードを選択します。
<b>GET SAME</b>	現在処理中のレコードを再読み取りします。
<b>HISTOGRAM</b>	データベースフィールドの値を読み取ります。
<b>IF</b>	論理条件に基づいてステートメントを実行します。

## 機能別ステートメント

---

<b>IF SELECTION</b>	一連の英数字フィールドのうちの1つにのみ値が含まれていることを確認します。
<b>ON ERROR</b>	ランタイムエラーをインターセプトします。インターセプトしない場合は、Natural エラーメッセージが表示された後、Natural プログラムが終了します。
<b>READ</b>	データベースファイル内の物理的または論理的な一連のレコードを読み取ります。
<b>READ WORK FILE</b>	ワークファイルからデータを読み取ります。
<b>REPEAT</b>	処理ループを開始し、指定された条件に基づいて終了します。
<b>SORT</b>	レコードをソートします。
<b>STORE</b>	新しいレコードをデータベースに追加します。
<b>UPDATE</b>	データベース内のレコードを更新します。
<b>UPLOAD PC FILE</b>	PC からメインフレーム、UNIX、または OpenVMS プラットフォームへのデータ転送を可能にします。

## Predict Case および Entire DB で使用可能なステートメント

---

次の Natural ステートメントは、Predict Case および Entire DB Engine と組み合わせて使用できます。

- DLOGOFF/DLOGON
- IMPORT
- EXPORT
- SHOW

これらのステートメントの詳細については、Predict Case ドキュメントを参照してください。

# 5 ACCEPT/REJECT

---

■ 機能 .....	22
■ 構文説明 .....	22
■ 複数の ACCEPT/REJECTステートメントの処理 .....	23
■ リミット表記 .....	23
■ ホールド状態 .....	23
■ 例 .....	24

```
{ ACCEPT }
{ REJECT } [IF] logical-condition
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [HISTOGRAM](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION DATA](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

ステートメント ACCEPT および REJECT は、ユーザー指定の論理条件を基準にして、レコードを受け入れまたは除外します。ACCEPT/REJECT ステートメントは、データレコードを読み込むステートメント ([FIND](#)、[READ](#)、[HISTOGRAM](#)、[CALL FILE](#)、[SORT](#)、[READ WORK FILE](#)) とともに、処理ループ内で使用します。指定した条件は、レコードが選択され、読み込まれた後に評価されます。

ACCEPT/REJECT ステートメントの処理に達すると、レコードを読むステートメントで開始された現在稼働している最も内側の処理ループが、内部的に参照されます。

ACCEPT/REJECT ステートメントがサブルーチン内にあり、それがレコードの除外 (REJECT) である場合、処理ループに入ったサブルーチンは自動的に終了します。そして、現在稼働している最も内側の処理ループの次のレコードで処理が続けられます。

## 構文説明

IF	<a href="#">FIND</a> 、 <a href="#">READ</a> 、または <a href="#">HISTOGRAM</a> ステートメントでレコードが選択され、読み込まれる場合、IF 節を ACCEPT または REJECT ステートメントで使用して、指定された条件に加えて論理条件の基準を指定することもできます。論理条件の基準は、レコードが読み込まれ、レコードの処理が開始した後に評価されます。
<i>logical-condition</i>	<p>基本の条件は、1つの関係式です。複数の関係式を論理演算子 (AND、OR) と組み合わせ、複合条件を構成することができます。</p> <p>また、演算式を使用して、1つの関係式を構成することもできます。</p> <p>論理条件に使用するフィールドとして、データベースフィールドあるいはユーザー定義変数が指定できます。論理条件の詳細については、『<a href="#">プログラミングガイド</a>』の「<a href="#">論理条件基準</a>」を参照してください。</p>



	ACCEPT/REJECT ステートメントを <b>HISTOGRAM</b> ステートメントで使用する場合、 <b>HISTOGRAM</b> ステートメントで指定したデータベースフィールドだけが論理条件として使用できます。
--	---

## 複数の ACCEPT/REJECT ステートメントの処理

通常1つの処理ループ内では、ACCEPT または REJECT ステートメントを使用します。2つ以上の ACCEPT/REJECT を連続して指定した場合は、次の条件が適用されます。

- 同一処理ループ内に、続けて ACCEPT および REJECT ステートメントを使用すると、指定した順序で処理されます。
- ACCEPT 条件を満たすと、そのレコードは受け入れ (ACCEPT) られます。同じ処理ループ内の連続した ACCEPT/REJECT ステートメントは処理されません。
- REJECT 条件を満たすと、そのレコードは除外 (REJECT) されます。同じ処理ループ内の連続した ACCEPT/REJECT ステートメントは処理されません。
- 処理が最後の ACCEPT/REJECT ステートメントに達すると、そのレコードを受け入れるかどうかは、最後の ACCEPT/REJECT ステートメントで決定されます。

他のステートメントが、複数の ACCEPT/REJECT ステートメントの間に挿入されていると、各 ACCEPT/REJECT ステートメントは独立して処理されます。

## リミット表記

ACCEPT または REJECT ステートメントを含む処理ループに対して、**LIMIT** ステートメントや他の制限指定が与えられている場合は、各処理レコードは受け入れられる (ACCEPT) か除外される (REJECT) かにかかわらずカウントされ、制限数と比較されます。

## ホールド状態

ACCEPT/REJECT 処理では、プロファイルパラメータ RI (ISNの解放) が RI=ON に設定されていない限り、ホールド中のレコードがホールド状態から解放されることはありません。

## 例

- 例 1 - ACCEPT
- 例 2 - ACCEPT / REJECT

### 例 1 - ACCEPT

```
** Example 'ACREX1': ACCEPT
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 SEX
  2 MAR-STAT
END-DEFINE
*
LIMIT 50
READ EMPLOY-VIEW
  ACCEPT IF SEX='M' AND MAR-STAT = 'S'
  WRITE NOTITLE '=' NAME '=' SEX 5X '=' MAR-STAT
END-READ
END
```

プログラム **ACREX1** の出力：

```
NAME: MORENO           S E X: M      MARITAL STATUS: S
NAME: VAUZELLE         S E X: M      MARITAL STATUS: S
NAME: BAILLET          S E X: M      MARITAL STATUS: S
NAME: HEURTEBISE      S E X: M      MARITAL STATUS: S
NAME: LION             S E X: M      MARITAL STATUS: S
NAME: DEZELUS          S E X: M      MARITAL STATUS: S
NAME: BOYER            S E X: M      MARITAL STATUS: S
NAME: BROUSSE          S E X: M      MARITAL STATUS: S
NAME: DROMARD          S E X: M      MARITAL STATUS: S
NAME: DUC              S E X: M      MARITAL STATUS: S
NAME: BEGUERIE        S E X: M      MARITAL STATUS: S
NAME: FOREST           S E X: M      MARITAL STATUS: S
NAME: GEORGES          S E X: M      MARITAL STATUS: S
```

## 例 2 - ACCEPT / REJECT

```

** Example 'ACREX2': ACCEPT/REJECT
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 SALARY      (1)
*
1 #PROC-COUNT (N8) INIT <0>
END-DEFINE
*
EMP. FIND EMPLOY-VIEW WITH NAME = 'JACKSON'
  WRITE NOTITLE *COUNTER NAME FIRST-NAME 'SALARY:' SALARY(1)
  /*
  ACCEPT IF SALARY (1) LT 50000
  WRITE *COUNTER 'ACCEPTED FOR FURTHER PROCESSING'
  /*
  REJECT IF SALARY (1) GT 30000
  WRITE *COUNTER 'NOT REJECTED'
  /*
  ADD 1 TO #PROC-COUNT
END-FIND
*
SKIP 2
WRITE NOTITLE 'TOTAL PERSONS FOUND ' *NUMBER (EMP.) /
              'TOTAL PERSONS SELECTED' #PROC-COUNT
END

```

プログラム **ACREX2** の出力：

```

1 JACKSON          CLAUDE          SALARY:      33000
1 ACCEPTED FOR FURTHER PROCESSING
2 JACKSON          FORTUNA          SALARY:      36000
2 ACCEPTED FOR FURTHER PROCESSING
3 JACKSON          CHARLIE          SALARY:      23000
3 ACCEPTED FOR FURTHER PROCESSING
3 NOT REJECTED

TOTAL PERSONS FOUND          3
TOTAL PERSONS SELECTED      1

```



# 6 ADD

---

▪ 機能 .....	28
▪ 構文説明 .....	28
▪ 例 .....	30

関連ステートメント：[COMPRESS](#) | [COMPUTE](#) | [DIVIDE](#) | [EXAMINE](#) | [MOVE](#) | [MOVE ALL](#) | [MULTIPLY](#) | [RESET](#) | [SEPARATE](#) | [SUBTRACT](#)

関連機能グループ：「[算術演算とデータ移動操作](#)」

## 機能

ADD ステートメントを使用して、複数のオペランドを追加します。



### 注意:

1. ADD ステートメント実行時、算術演算に使用される各オペランドは、有効な値を持っている必要があります。
2. 配列を含む加算については、「[配列の算術演算](#)」を参照してください。
3. オペランドのフォーマットについては、「[フォーマット混合式のパフォーマンスについて](#)」を参照してください。

## 構文説明

このステートメントには、2つの異なる構造が可能です。

- [構文 1](#)
- [構文 2](#)

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

### 構文 1

```
ADD[ROUNDED] operand1... T0 operand2
```

オペランド定義テーブル（構文 1）：

オペランド	構文要素				フォーマット								ステートメント参照	ダイナミック定義				
<i>operand1</i>	C	S	A	N			N	P	I	F	D	T					可	不可
<i>operand2</i>		S	A	M			N	P	I	F	D	T					可	可

構文要素の説明：

<i>operand1</i>	<i>operand1</i> は加数です。
<b>ROUNDED</b>	キーワード <b>ROUNDED</b> を使用すると、結果が四捨五入されます。四捨五入のルールについては、「演算割り当てのルール」を参照してください。
<b>TO</b> <i>operand2</i>	<i>operand2</i> は加算に含められ、演算結果を受け取ります。

例：

ステートメント

```
ADD #A(*) TO #B(*) is equivalent to COMPUTE #B(*) := #A(*) + #B(*)
ADD #S TO #R is equivalent to COMPUTE #R := #S + #R
ADD #S #T TO #R is equivalent to COMPUTE #R := #S + #T + #R
ADD #A(*) TO #R is equivalent to COMPUTE #R := #A(*) + #R
```

## 構文 2

`ADD[ROUNDED] operand1... GIVING operand2`


オペランド定義テーブル（構文 2）：

オペランド	構文要素				フォーマット										ステートメント参照	ダイナミック定義				
<i>operand1</i>	C	S	A	N				N	P	I	F		D	T					可	不可
<i>operand2</i>		S	A	M	A	U	N	P	I	F	B*	D	T					可	可	

\**operand3* のフォーマット B は、4 以下の長さでのみ使用できます。

構文要素の説明：

<i>operand1</i>	<i>operand1</i> は加数です。
<b>ROUNDED</b>	キーワード <b>ROUNDED</b> を使用すると、結果が四捨五入されます。四捨五入のルールについては、「演算割り当てのルール」を参照してください。
<b>GIVING</b> <i>operand2</i>	<i>operand2</i> は演算結果を受け取るのみで、加算の対象にはなりません。  <i>operand2</i> を英数字フォーマットで定義した場合、結果は英数字フォーマットに変換されます。

 **注意:** 構文 2 を使用する場合、次のことが適用されます。キーワード **GIVING** の左側の (*operand1*) フィールドのみが加算の項であり、キーワード **GIVING** (*operand2*) の右側の

フィールドは結果値を受け取るために使用されるだけです。単一の (*operand1*) フィールドのみが指定されている場合、ADD 演算は割り当てに変わります。

例：

ステートメント

```
ADD #S      GIVING #R  is equivalent to  COMPUTE #R := #S
ADD #S #T   GIVING #R  is equivalent to  COMPUTE #R := #S + #T
ADD #A(*) 0  GIVING #R  is equivalent to  COMPUTE #R := #A(*) + 0
           which is a legal operation, due to the rules defined in Arithmetic Operations with Arrays
ADD #A(*)   GIVING #R  is equivalent to  COMPUTE #R := #A(*)
           which is an illegal operation, due to the rules defined in Assignment Operations with Arrays
```

## 例

```
** Example 'ADDEX1': ADD
*****
DEFINE DATA LOCAL
1 #A      (P2)
1 #B      (P1.1)
1 #C      (P1)
1 #DATE   (D)
1 #ARRAY1 (P5/1:4,1:4) INIT (2,*) <5>
1 #ARRAY2 (P5/1:4,1:4) INIT (4,*) <10>
END-DEFINE
*
ADD +5 -2 -1 GIVING #A
WRITE NOTITLE 'ADD +5 -2 -1 GIVING #A' 15X '=' #A
*
ADD .231 3.6 GIVING #B
WRITE          / 'ADD .231 3.6 GIVING #B' 15X '=' #B
*
ADD ROUNDED 2.9 3.8 GIVING #C
WRITE          / 'ADD ROUNDED 2.9 3.8 GIVING #C' 8X '=' #C
*
MOVE *DATX TO #DATE
ADD 7 TO #DATE
WRITE          / 'CURRENT DATE:'          *DATX (DF=L) 13X
                'CURRENT DATE + 7:' #DATE (DF=L)
*
WRITE          / '#ARRAY1 AND #ARRAY2 BEFORE ADDITION'
                / '=' #ARRAY1 (2,*) '=' #ARRAY2 (4,*)
ADD #ARRAY1 (2,*) TO #ARRAY2 (4,*)
WRITE          / '#ARRAY1 AND #ARRAY2 AFTER ADDITION'
                / '=' #ARRAY1 (2,*) '=' #ARRAY2 (4,*)
```



```
*  
END
```

プログラム **ADDEX1** の出力：

```
ADD +5 -2 -1 GIVING #A          #A:  2  
ADD .231 3.6 GIVING #B          #B:  3.8  
ADD ROUNDED 2.9 3.8 GIVING #C    #C:  7  
CURRENT DATE: 2005-01-10        CURRENT DATE + 7: 2005-01-17  
  
#ARRAY1 AND #ARRAY2 BEFORE ADDITION  
#ARRAY1:      5      5      5      5 #ARRAY2:      10      10      10      10  
  
#ARRAY1 AND #ARRAY2 AFTER ADDITION  
#ARRAY1:      5      5      5      5 #ARRAY2:      15      15      15      15
```



# 7 ASSIGN

---

COMPUTE ステートメントを参照してください。



# 8 AT BREAK

---

▪ 機能 .....	36
▪ 構文説明 .....	37
▪ 複数ブレイクレベルの階層 .....	38
▪ 例 .....	39

## ストラクチャードモード構文

```
[AT] BREAK [(r)] [OF] operand1 [/n/]  
statement ...  
END-BREAK
```

## レポーティングモード構文

```
[AT] BREAK [(r)] [OF] operand1 [/n/]  
    { statement  
      DO statement...DOEND }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION DATA](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

---

AT BREAK ステートメントは、[コントロールフィールド](#)の値が変わるたびに、1つ以上のステートメントを実行します。このステートメントは、自動ブレイク処理と組み合わせて使用します。また、[FIND](#)、[READ](#)、[HISTOGRAM](#)、[SORT](#)、[READ WORK FILE](#) ステートメントとともに使用します。

自動ブレイク処理は、次のように実行されます。処理ループによりレコードが読み取られた直後に、コントロールフィールドがチェックされます。前のレコードと比較して値が変更されていることが検出された場合、AT BREAK ステートメントブロックに含まれるステートメントが実行されます。これは、処理ループ内の最初のレコードには適用されません。さらに、（レコードの読み取りが完了したとき、または [ESCAPE BOTTOM](#) ステートメントによって）処理ループが終了した場合、AT BREAK ステートメントブロック内のステートメントの最終実行がトリガされます。

詳細については、『[プログラミングガイド](#)』の「[自動ブレイク処理](#)」を参照してください。

AT BREAK ステートメントブロックは、ブレイク条件の発生時にそのステートメントを含んだオブジェクトがアクティブな場合にだけ実行されます。

AT BREAK 条件内で新しく処理ループを開始することができます。ただし、このループは、その AT BREAK 条件内で閉じる必要があります。

このステートメントは非手続き型なので、プログラム内の位置ではなくイベントによって実行されます。

AT BREAK ステートメントで使用できる Natural システム関数については、『システム関数』ドキュメントの「処理ループで使用する Natural システム関数」および『プログラミングガイド』の「AT BREAK ステートメントとシステム関数の例」を参照してください。

詳細については、『プログラミングガイド』の「AT BREAK ステートメント」も参照してください。このセクションでは、次のトピックについて説明しています。

- データベースフィールドに基づくコントロールブレイク
- ユーザー定義変数に基づくコントロールブレイク

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	動的定義
<i>operand1</i>	S	A U N P I F B D T L	可	不可

構文要素の説明：

(r)	<p>参照表記： デフォルトで、最終の AT BREAK 条件（ループ終了処理）は、常に <b>FIND</b>、<b>READ</b>、<b>READ WORK FILE</b>、<b>HISTOGRAM</b>、または <b>SORT</b> ステートメントで開始された、一番外側で稼働している処理ループに関連します。</p> <p>表記 (r) を使用すると、AT BREAK ステートメントの最終ブレイクを、現在オープンしている特定の処理ループに関連付けることができます（AT BREAK ステートメントが含まれるそのループを含めて外側のどのループにでも関連付けられます）。</p> <p>例：</p> <pre> ... READ ...   FIND ...     FIND ...       AT BREAK ...         FIND ...           END-FIND         END-BREAK       END-FIND     END-FIND   END-READ ... </pre>
-----	---

	<p>この例では、最終の AT BREAK 条件は、行番号 0120 で開始する READ ループに関連付けられます。また、AT BREAK 条件を行番号 0130 や 0140 で開始する FIND ループに関連付けられます（ただし、行番号 0160 の FIND には関連付けられません）。</p> <p>ブレイク階層に (r) を指定する場合は、最初の AT BREAK ステートメントで指定する必要があります。この指定は、その後の全 AT BREAK ステートメントにも適用されます。</p>
<i>operand1</i>	<p>コントロールフィールド： ブレイクコントロールフィールドとして使用されるフィールドで、通常はデータベースフィールドです。ユーザー定義変数を使用する場合は、自動ブレイク処理の評価の前に初期化されている必要があります（BEFORE BREAK PROCESSING ステートメントを参照）。配列の特定のオカレンスをコントロールフィールドとして使用することもできます。</p>
<i>lnl</i>	<p>-表記 /n/ を使用すると、値の変化を調べるためにチェックされるのは、コントロールフィールドの（左から右へ数えて）最初の n 個の位置だけであることを示すことができます。この表記は、オペランドがフォーマット A、B、N、または P の場合に使用できます。</p> <p>コントロールブレイクは、コントロールフィールドの値が変わるか、または AT BREAK ステートメントを適用する処理ループですべてのレコードの処理が終了したときに発生します。</p>
<b>END-BREAK</b>	<p>Natural 予約語 END-BREAK を使用して、AT BREAK ステートメントを終了させる必要があります。</p>

## 複数ブレイクレベルの階層

同一プログラムモジュール内の同一処理ループに、複数の AT BREAK ステートメントを指定することができます。同一処理ループに指定した複数の AT BREAK ステートメントは、連続指定しても、間に他のステートメントを指定しても、ブレイクレベルの階層を形成します。最初の AT BREAK ステートメントは、一番低いレベルのコントロールブレイクです。続く各 AT BREAK ステートメントは、コントロールブレイクの階層でそれぞれ次に高いブレイクレベルとなります。

階層構造を作る処理ループでも、それぞれのループ内に AT BREAK ステートメントの階層を持つことができます。

例：

ストラクチャードモード：	レポートイングモード：
FIND ...	FIND ...
AT BREAK	AT BREAK
...	DO
END-BREAK	...
AT BREAK	DOEND
...	AT BREAK
END-BREAK	DO
AT BREAK	...
...	DOEND
END-BREAK	...
END-FIND	



ストラクチャードモード：	レポートモード：
...	

あるブレイクレベルのコントロールフィールドの値が変わると、そのブレイクレベルと、より低いブレイクレベルの処理がすべて行われます（より低いブレイクレベルのコントロールフィールドの値とは関係なく行われます）。

複数のブレイク処理を連続して指定すると、プログラムメンテナンスが容易になります。

下記の「[例3](#)」および『プログラミングガイド』の「[複数のコントロールブレイクレベル](#)」セクションを参照してください。

## 例

このセクションでは、次のトピックについて説明します。

- [例 1 - AT BREAK](#)
- [例 2 - /n/ 表記を使用した AT BREAK](#)
- [例 3 - AT BREAK による複数ブレイクレベルの指定](#)

AT BREAK の他の例については、「[処理ループで使用する Natural システム関数](#)」、および例 [ATBEX3](#) と [ATBEX4](#) を参照してください。

### 例 1 - AT BREAK

```

** Example 'ATBEX1S': AT BREAK (structured mode)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 COUNTRY
  2 NAME
END-DEFINE
*
LIMIT 10
READ EMPLOY-VIEW BY CITY
AT BREAK OF CITY
  SKIP 1
  END-BREAK
  DISPLAY NOTITLE CITY (IS=ON) COUNTRY (IS=ON) NAME
END-READ
*
END

```

プログラム **ATBEX1S** の出力：

CITY	COUNTRY	NAME
AIKEN	USA	SENKO
AIX EN OTHE	F	GODEFROY
AJACCIO		CANALE
ALBERTSLUND	DK	PLOUG
ALBUQUERQUE	USA	HAMMOND ROLLING FREEMAN LINCOLN
ALFRETON	UK	GOLDBERG
ALICANTE	E	GOMEZ

レポートモードの例については、ライブラリ SYSEXRM のプログラム **ATBEX1R** を参照してください。

## 例 2 - /n/ 表記を使用した AT BREAK

```
** Example 'ATBEX2': AT BREAK (with /n/ notation)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 DEPT
  2 NAME
END-DEFINE
*
LIMIT 10
READ EMPLOY-VIEW BY DEPT STARTING FROM 'A'
AT BREAK OF DEPT /4/
  SKIP 1
  END-BREAK
  DISPLAY NOTITLE DEPT NAME
END-READ
*
END
```

プログラム **ATBEX2** の出力：

```
DEPARTMENT      NAME
  CODE
-----
ADMA01      JENSEN
ADMA01      PETERSEN
ADMA01      MORTENSEN
ADMA01      MADSEN
ADMA01      BUHL
ADMA02      HERMANSEN
ADMA02      PLOUG
ADMA02      HANSEN

COMP01      HEURTEBISE
COMP01      TANCHOU
```

### 例 3 - AT BREAK による複数ブレイクレベルの指定

```
** Example 'ATBEX5S': AT BREAK (multiple break levels) (structured mode)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 DEPT
  2 NAME
  2 LEAVE-DUE
1 #LEAVE-DUE-L (N4)
END-DEFINE
*
LIMIT 5
FIND EMPLOY-VIEW WITH CITY = 'PHILADELPHIA' OR = 'PITTSBURGH'
      SORTED BY CITY DEPT
  MOVE LEAVE-DUE TO #LEAVE-DUE-L
  DISPLAY CITY (IS=ON) DEPT (IS=ON) NAME #LEAVE-DUE-L
/*
  AT BREAK OF DEPT
    WRITE NOTITLE /
      T*DEPT OLD(DEPT) T*#LEAVE-DUE-L SUM(#LEAVE-DUE-L) /
  END-BREAK
  AT BREAK OF CITY
    WRITE NOTITLE
      T*CITY OLD(CITY) T*#LEAVE-DUE-L SUM(#LEAVE-DUE-L) //
  END-BREAK
END-FIND
*
END
```

プログラム **ATBEX5** の出力：

CITY	DEPARTMENT CODE	NAME	#LEAVE-DUE-L
PHILADELPHIA	MGMT30	WOLF-TERROINE	11
		MACKARNESS	27
	MGMT30		38
	TECH10	BUSH	39
		NETTLEFOLDS	24
TECH10		63	
PHILADELPHIA			101
PITTSBURGH	MGMT10	FLETCHER	34
	MGMT10		34
PITTSBURGH			34

レポートモードの例については、ライブラリ SYSEXRM のプログラム **ATBEX5R** を参照してください。

# 9 AT END OF DATA

---

▪ 機能 .....	44
▪ 制限事項 .....	45
▪ 構文説明 .....	45
▪ 例 .....	46

ストラクチャードモード構文

```
[AT]END [OF] DATA [(r)]  
statement ...  
END-ENDDATA
```

レポートイングモード構文

```
[AT]END [OF] DATA [(r)]  
{  
    statement  
    DO statement ... DOEND  
}
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION DATA](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

---

AT END OF DATA ステートメントは、データベース処理ループで選択された全レコードの処理が終了したときに行う処理を指定します。

このセクションでは、次のトピックについて説明します。

- [処理](#)
- [データベースフィールドの値](#)
- [位置](#)
- [システム関数](#)

『[プログラミングガイド](#)』の「[AT START/END OF DATA ステートメント](#)」を参照してください。

## 処理

このステートメントは非手続き型なので、プログラム内の位置ではなくイベントによって実行されます。

## データベースフィールドの値

処理ループで AT END OF DATA 条件が発生したときに、すべてのデータベースフィールドは、最後に処理されたレコードの値になります。

## 位置

このステートメントは、対応するループ生成ステートメントと同じプログラムモジュール内に指定する必要があります。

## システム関数

AT END OF DATA ステートメントで使用できる Natural システム関数については、『システム関数』ドキュメントの「処理ループでのシステム関数の使用」セクションを参照してください。

## 制限事項

- このステートメントは、**FIND**、**READ**、**READ WORK FILE**、**HISTOGRAM**、または **SORT** ステートメントで始まる処理ループだけに使用できます。
- また、処理ループ内で 1 回だけ使用できます。
- END OF DATA が参照する処理ループに入らなかった場合、このステートメントは評価されません。

## 構文説明

(r)	特定の処理ループの参照： AT END OF DATA ステートメントは、表記 (r) によって、アクティブな特定の処理ループに関連付けることができます。この表記がない場合、AT END OF DATA ステートメントはアクティブなデータベース処理ループ内の一番外側に関連付けられます。
<b>END-ENDDATA</b>	Natural 予約語 END-ENDDATA を使用して、AT END OF DATA ステートメントを終了させる必要があります。

## 例

```

** Example 'AEDEX1S': AT END OF DATA
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
  2 SALARY (1)
  2 CURR-CODE (1)
END-DEFINE
*
LIMIT 5
EMP. FIND EMPLOY-VIEW WITH CITY = 'STUTTGART'
  IF NO RECORDS FOUND
    ENTER
  END-NOREC
  DISPLAY PERSONNEL-ID NAME FIRST-NAME
    SALARY (1) CURR-CODE (1)
/*
  AT END OF DATA
  IF *COUNTER (EMP.) = 0
    WRITE 'NO RECORDS FOUND'
    ESCAPE BOTTOM
  END-IF
  WRITE NOTITLE / 'SALARY STATISTICS:'
    / 7X 'MAXIMUM:' MAX(SALARY(1)) CURR-CODE (1)
    / 7X 'MINIMUM:' MIN(SALARY(1)) CURR-CODE (1)
    / 7X 'AVERAGE:' AVER(SALARY(1)) CURR-CODE (1)
  END-ENDDATA
/*
END-FIND
*
END

```

「処理ループで使用する *Natural* システム関数」も参照してください。

プログラム **AEDEX1S** の出力：

PERSONNEL ID	NAME	FIRST-NAME	ANNUAL SALARY	CURRENCY CODE
11100328	BERGHAUS	ROSE	70800	DM
11100329	BARTHEL	PETER	42000	DM
11300313	AECKERLE	SUSANNE	55200	DM
11300316	KANTE	GABRIELE	61200	DM
11500304	KLUGE	ELKE	49200	DM



```
SALARY STATISTICS:  
  MAXIMUM:      70800 DM  
  MINIMUM:      42000 DM  
  AVERAGE:      55680 DM
```

レポートモードの例については、ライブラリ SYSEXRM のプログラム [AEDEX1R](#) を参照してください。



# 10 AT END OF PAGE

---

▪ 機能 .....	50
▪ 構文説明 .....	52
▪ 例 .....	52

## ストラクチャードモード構文

```
[AT] END [OF] PAGE [(rep)]
statement ...
END-ENDPAGE
```

## レポートニングモード構文

```
[AT] END [OF] PAGE [(rep)]
{
    statement
    DO statement ... DOEND
}
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT TOP OF PAGE](#) | [CLOSE PRINTER](#) | [DEFINE PRINTER](#) | [DISPLAY](#) | [EJECT](#) | [FORMAT](#) | [NEWPAGE](#) | [PRINT](#) | [SKIP](#) | [SUSPEND IDENTICAL SUPPRESS](#) | [WRITE](#) | [WRITE TITLE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

---

AT END OF PAGE ステートメントは、END OF PAGE（ページ終了）条件が発生したときに行う処理を指定します（『[パラメータリファレンス](#)』ドキュメントのセッションパラメータ PS を参照）。END OF PAGE 条件は、[SKIP](#) または [NEWPAGE](#) ステートメントでも発生します。ただし、[EJECT](#) または [INPUT](#) ステートメントの場合は発生しません。

『[プログラミングガイド](#)』の次のセクションも参照してください。

- [データ出力制御](#)
- [レポート指定 - \(rep\) 表記](#)
- [出力ページのレイアウト](#)
- [AT END OF PAGE ステートメント](#)

## 処理

AT END OF PAGE ステートメントブロックは、END OF PAGE 条件が発生したときにそのステートメントブロックを含むオブジェクトが、アクティブな場合にのみ実行されます。

AT END OF PAGE ステートメントは、内部サブルーチン内には指定できません。

このステートメントは非手続き型なので、プログラム内の位置ではなくイベントによって実行されます。

## 論理ページサイズ

END OF PAGE のチェックは、DISPLAY や WRITE ステートメントの処理の完了後に実行されます。そのため、DISPLAY や WRITE ステートメントで複数行の出力を行うと、END OF PAGE 条件が発生する前に、物理ページのオーバーフローを起こすことがあります。

AT END OF PAGE ステートメントでプリントされる情報が、タイトルと同じ物理ページに出力されることを保証するには、物理的なページサイズより小さい論理ページサイズ（セッションパラメータ PS）を指定する必要があります。

## 最終ページの扱い

メインプログラム内では、ESCAPE、STOP、または END によりメインプログラムの実行が終了すると、END OF PAGE 条件が発生します。

サブルーチン内では、ESCAPE-ROUTINE、RETURN、または END-SUBROUTINE によりサブルーチンの実行が終了すると、END OF PAGE 条件が発生します。

## システム関数

AT END OF PAGE ステートメントで使用できる Natural システム関数については、『システム関数』ドキュメントの「処理ループでのシステム関数の使用」セクションを参照してください。

AT END OF PAGE ステートメントブロック内で、システム関数を使用する場合は、GIVE SYSTEM FUNCTIONS 節を、対応する DISPLAY ステートメントで指定する必要があります。

## AT END OF PAGE と INPUT ステートメント

AT END OF PAGE ステートメントブロック内に、INPUT ステートメントを指定すると、NEWPAGE（ページ換え）処理は実行されません。INPUT ステートメントで作成した行を同じ物理ページに表示するために、セッションパラメータ PS でページサイズを物理ページサイズより小さい値に指定する必要があります。

以下の項目も参照してください。

- INPUT ステートメントの画面分割機能

■ 例2 - INPUT ステートメントでの AT END OF PAGE

## 構文説明

(rep)	<p>レポート指定：                  表記 (rep) を使用すると、AT END OF PAGE ステートメントを適用するレポートの ID を指定できます。範囲 0~31 の値、または <b>DEFINE PRINTER</b> ステートメントを使用して割り当てた論理名を指定できます。</p> <p>(rep) を指定しない場合、AT END OF PAGE ステートメントは最初のレポート (レポート 0) に適用されます。</p> <p>Natural で作成した出力レポートのフォーマットを制御する方法については、『プログラミングガイド』の「データ出力の制御」を参照してください。</p>
END-ENDPAGE	<p>Natural 予約語 END-ENDPAGE を使用して、AT END OF PAGE ステートメントを終了させる必要があります。</p>

## 例

- 例 1 - AT END OF PAGE
- 例 2 - INPUT ステートメントでの AT END OF PAGE

### 例 1 - AT END OF PAGE

```

** Example 'AEPEX1S': AT END OF PAGE (structured mode)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 JOB-TITLE
  2 SALARY (1)
  2 CURR-CODE (1)
END-DEFINE
*
FORMAT PS=10
LIMIT 10
READ EMPLOY-VIEW BY PERSONNEL-ID FROM '20017000'
  DISPLAY NOTITLE GIVE SYSTEM FUNCTIONS
    NAME JOB-TITLE 'SALARY' SALARY(1) CURR-CODE (1)
  /*

AT END OF PAGE
WRITE / 28T 'AVERAGE SALARY: ...' AVER(SALARY(1)) CURR-CODE (1)
    
```

```
END-ENDPAGE
```

```
END-READ
```

```
*
```

```
END
```

「処理ループで使用する *Natural* システム関数」も参照してください。

プログラム **AEPEX1S** の出力：

NAME	CURRENT POSITION	SALARY	CURRENCY CODE
CREMER	ANALYST	34000	USD
MARKUSH	TRAINEE	22000	USD
GEE	MANAGER	39500	USD
KUNEY	DBA	40200	USD
NEEDHAM	PROGRAMMER	32500	USD
JACKSON	PROGRAMMER	33000	USD
AVERAGE SALARY: ...		33533	USD

レポートモードの例については、ライブラリ SYSEXRM のプログラム **AEPEX1R** を参照してください。

## 例 2 - INPUT ステートメントでの AT END OF PAGE

```
** Example 'AEPEX2': AT END OF PAGE (with INPUT)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 POST-CODE
  2 CITY
*
1 #START-NAME (A20)
END-DEFINE
*
FORMAT PS=21
*
REPEAT
  READ (15) EMPLOY-VIEW BY NAME = #START-NAME
  DISPLAY NOTITLE NAME FIRST-NAME POST-CODE CITY
  END-READ
  NEWPAGE
/*
AT END OF PAGE
```

```

MOVE NAME TO #START-NAME
INPUT / '-' (79)
      / 10T 'Reposition to name ==>'
          #START-NAME (AD=MI) '('.'.' to exit)'
IF #START-NAME = '.'
  STOP
END-IF
END-ENDPAGE
/*
END-REPEAT
END

```

プログラム **AEPEX2S** の出力：

NAME	FIRST-NAME	POSTAL ADDRESS	CITY
ABELLAN	KEPA	28014	MADRID
ACHIESON	ROBERT	DE3 4TR	DERBY
ADAM	SIMONE	89300	JOIGNY
ADKINSON	JEFF	11201	BROOKLYN
ADKINSON	PHYLLIS	90211	BEVERLEY HILLS
ADKINSON	HAZEL	20760	GAITHERSBURG
ADKINSON	DAVID	27514	CHAPEL HILL
ADKINSON	CHARLIE	21730	LEXINGTON
ADKINSON	MARTHA	17010	FRAMINGHAM
ADKINSON	TIMMIE	17300	BEDFORD
ADKINSON	BOB	66044	LAWRENCE
AECKERLE	SUSANNE	7000	STUTTGART
AFANASSIEV	PHILIP	39401	HATTIESBURG
AFANASSIEV	ROSE	60201	EVANSTON
AHL	FLEMMING	2300	SUNDBY

-----

Reposition to name ==> AHL ('.' to exit)



# 11 AT START OF DATA

---

▪ 機能 .....	56
▪ 構文説明 .....	57
▪ 例 .....	57

ストラクチャードモード構文

```
[AT] START [OF] DATA [(r)]  
statement ...  
END-START
```

レポートイングモード構文

```
[AT] START [OF] DATA [(r)]  
{  
    statement  
    DO statement... DOEND  
}
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION DATA](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

---

AT START OF DATA ステートメントは、[READ](#)、[FIND](#)、[HISTOGRAM](#)、[SORT](#)、または [READ WORK FILE](#) ステートメントで始まる処理ループで最初のレコードが読まれた直後の処理を指定します。

『[プログラミングガイド](#)』の「[AT START/END OF DATA ステートメント](#)」を参照してください。

### 処理

ループ開始ステートメントに WHERE 節がある場合は、基本検索条件と WHERE 条件の両方を満たす最初のレコードが読まれた時点で、AT START OF DATA 条件が真になります。

このステートメントは非手続き型なので、プログラム内の位置ではなくイベントによって実行されます。

## データベースフィールドの値

データベースフィールドには、AT START OF DATA 条件が真になるレコードの値が含まれます。つまり、処理するすべてのレコード集合の第 1 レコードの値になります。

## 位置

このステートメントは、処理ループの中に指定する必要があります。また、その処理ループに対して 1 回だけ使用できます。

## 構文説明

(r)	特定の処理ループの参照： AT START OF DATA ステートメントは、表記 (r) によって、アクティブな特定の処理ループに関連付けることができます。この表記を使用しなかった場合、ステートメントはアクティブな処理ループ内の一番外側に関連付けられます。
END-START	Natural 予約語 END-START を使用して、AT START OF DATA ステートメントを終了させる必要があります。

## 例

```

** Example 'ASDEX1S': AT START OF DATA (structured mode)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 CITY
*
1 #CNTL (A1) INIT <' '>
1 #CITY (A20) INIT <' '>
END-DEFINE
*
REPEAT
  INPUT 'ENTER VALUE FOR CITY' #CITY
  IF #CITY = ' ' OR = 'END'
    STOP
  END-IF
  FIND EMPLOY-VIEW WITH CITY = #CITY
  IF NO RECORDS FOUND
    WRITE NOTITLE NOHDR 'NO RECORDS FOUND'
    ESCAPE BOTTOM
  END-NOREC

```

```
/*
AT START OF DATA
  INPUT (AD=0) 'RECORDS FOUND' *NUMBER //
              'ENTER 'D' TO DISPLAY RECORDS' #CNTL (AD=A)
  IF #CNTL NE 'D'
    ESCAPE BOTTOM
  END-IF
END-START
/*
  DISPLAY NAME FIRST-NAME
END-FIND
END-REPEAT
END
```

プログラム **ASDEX1S** の出力：

```
ENTER VALUE FOR CITY PARIS
```

市町村名を入力して確認した後：

```
RECORDS FOUND      26
```

```
ENTER 'D' TO DISPLAY RECORDS D
```

表示されるレコード：

NAME	FIRST-NAME
MAIZIERE	ELISABETH
MARX	JEAN-MARIE
REIGNARD	JACQUELINE
RENAUD	MICHEL
REMOUE	GERMAINE
LAVENDA	SALOMON
BROUSSE	GUY
GIORDA	LOUIS
SIECA	FRANCOIS
CENSIER	BERNARD
DUC	JEAN-PAUL
CAHN	RAYMOND
MAZUY	ROBERT
FAURIE	HENRI
VALLY	ALAIN
BRETON	JEAN-MARIE
GIGLEUX	JACQUES
KORAB-BRZOZOWSKI	BOGDAN
XOLIN	CHRISTIAN
LEGRIS	ROGER
VVVV	

レポートモードの例については、ライブラリ SYSEXRM のプログラム **ASDEX1R** を参照してください。



# 12 AT TOP OF PAGE

---

▪ 機能 .....	62
▪ 制限事項 .....	63
▪ 構文説明 .....	63
▪ 例 .....	63

ストラクチャードモード構文

```
[AT] TOP [OF] PAGE [(rep)]
statement ...
END-TOPPAGE
```

レポートイングモード構文

```
[AT] TOP [OF] PAGE [(rep)]
{ statement
  DOstatement ... DOEND }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT END OF PAGE](#) | [CLOSE PRINTER](#) | [DEFINE PRINTER](#) | [DISPLAY](#) | [EJECT](#) | [FORMAT](#) | [NEWPAGE](#) | [PRINT](#) | [SKIP](#) | [SUSPEND IDENTICAL SUPPRESS](#) | [WRITE](#) | [WRITE TITLE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

---

AT TOP OF PAGE ステートメントは、新しいページが始まるときの処理を指定します。

『プログラミングガイド』の次のセクションも参照してください。

- データ出力制御
- レポート指定 - (rep) 表記
- 出力ページのレイアウト
- AT TOP OF PAGE ステートメント

## 処理

新しいページが始まるのは、内部の行カウンタが Natural セッションパラメータ PS (Natural レポートのページサイズ) で設定されたページサイズを超過したとき、あるいは [NEWPAGE](#) ステートメントが実行されたときです。どちらの場合も、AT 条件が真となります。[EJECT](#) ステートメントは、新しいページを開始しますが、AT 条件を発生させることはありません。

AT TOP OF PAGE ステートメントブロックは、AT 条件が発生したときに、そのステートメントブロックを含むオブジェクトがアクティブな場合にのみ実行されます。

AT TOP OF PAGE 処理の結果の出力は、タイトル行と 1 行の空行に続いて出力されます。



このステートメントは非手続き型なので、プログラム内の位置ではなくイベントによって実行されます。

## 制限事項

AT TOP OF PAGE ステートメントは、内部サブルーチン内には指定できません。

## 構文説明

<i>(rep)</i>	<p>レポート指定： 表記 (<i>rep</i>) を使用すると AT TOP OF PAGE ステートメントを提供するレポートの ID を指定できます。</p> <p>範囲 0~31 の値、または <b>DEFINE PRINTER</b> ステートメントを使用して割り当てた論理名を指定できます。</p> <p>(<i>rep</i>) を指定しない場合、AT TOP OF PAGE ステートメントは最初のレポート (レポート 0) に適用されます。</p> <p>Natural で作成した出力レポートのフォーマットを制御する方法については、『プログラミングガイド』の「データ出力の制御」を参照してください。</p>
<b>END-TOPPAGE</b>	Natural 予約語 END-TOPPAGE を使用して、AT TOP OF PAGE ステートメントを終了させる必要があります。

## 例

```

** Example 'ATPEX1S': AT TOP OF PAGE (structured mode)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 CITY
  2 DEPT
END-DEFINE
*
FORMAT PS=15
LIMIT 15
READ EMPLOY-VIEW BY NAME STARTING FROM 'L'
  DISPLAY 2X NAME 4X FIRST-NAME CITY DEPT
  WRITE TITLE UNDERLINED 'EMPLOYEE REPORT'
  WRITE TRAILER '-' (78)

```

```
/*
AT TOP OF PAGE
  WRITE 'BEGINNING NAME:' NAME
END-TOPPAGE
/*
AT END OF PAGE
  SKIP 1
  WRITE 'ENDING NAME:  ' NAME
END-ENDPAGE
END-READ
END
```

プログラム **ATPEX1S** の出力：

```
EMPLOYEE REPORT
-----
BEGINNING NAME: LAFON
                NAME                FIRST-NAME                CITY                DEPARTMENT
                NAME                CODE
-----
LAFON           CHRISTIANE                PARIS                VENT18
LANDMANN        HARRY                    ESCHBORN             MARK29
LANE            JACQUELINE               DERBY                MGMT02
LANKATILLEKE   LALITH                   FRANKFURT            PROD22
LANNON          BOB                      LINCOLN              SALE20
LANNON          LESLIE                   SEATTLE              SALE30
LARSEN          CARL                     FARUM                SYSA01
LARSEN          MOGENS                   VEMMELEV             SYSA02
-----
ENDING NAME:   LARSEN
```

レポートモードの例については、ライブラリ SYSEXRM のプログラム **ATPEX1R** を参照してください。

# 13 BACKOUT TRANSACTION

---

■ 機能 .....	66
■ 制限事項 .....	67
■ データベース固有の考慮事項 .....	67
■ 例 .....	67

**BACKOUT [TRANSACTION]**

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION DATA](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

---

## 機能

BACKOUT TRANSACTION ステートメントは、現在の論理トランザクション内で実行された、すべてのデータベース更新をバックアウトします。このステートメントも、トランザクション中に保持されていたすべてのレコードを解放します。

BACKOUT TRANSACTION ステートメントは、Natural の制御下でデータベーストランザクションが発生した場合にのみ実行されます。どのデータベースに対して、ステートメントが実行されるかは、Natural プロファイルパラメータ ET (END/BACKOUT TRANSACTION ステートメントの実行) の設定に依存します。

- ET=OFF の場合、トランザクションで影響を受けたデータベースに対してのみ、ステートメントが実行されます。
- ET=ON の場合、BACKOUT TRANSACTION または [END TRANSACTION](#) ステートメントの最後の実行以後に参照された全データベースに対して、ステートメントが実行されます。

### Natural が発行する BACKOUT TRANSACTION

ユーザーが端末コマンド (コマンド %% または CLEAR キー) で現在の Natural を中断した場合、Natural によって BACKOUT TRANSACTION ステートメントが発行されます。

詳細については、『[端末コマンド](#)』ドキュメントで端末コマンド %% を参照してください。

## 追加情報

トランザクションバックアウト機能の使用に関する詳細については、『プログラミングガイド』の「データベース更新-トランザクション処理」セクションおよび「トランザクションのバックアウト」セクションを参照してください。

## 制限事項

このステートメントは Entire System Server では使用できません。

## データベース固有の考慮事項

DLI データベース	PSB スケジューリングはシンクポイント要求によって終了するため、Natural では、BACKOUT TRANSACTION ステートメントの実行前にPSB ポジションを保存します。次のコマンドを実行する前に、Natural では PSB の再スケジュールを行い、PCB ポジションをバックアウト前のポジションに戻そうとします。バックアウトと次のコマンドの間にポイントされているセグメントが削除されていると、PCB ポジションが前方にシフトされていることがあります。
SQL データベース	ほとんどの SQL データベースは作業の論理ユニットが終了するときすべてのカーソルを閉じるため、BACKOUT TRANSACTION ステートメントをデータベースの更新処理ループ内に指定することはできません。このようなループの後に指定する必要があります。

## 例

```

** Example 'BOTEX1': BACKOUT TRANSACTION
**
** CAUTION: Executing this example will modify the database records!
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 DEPT
  2 LEAVE-DUE
  2 LEAVE-TAKEN
*
1 #DEPT (A6)
1 #RESP (A3)
END-DEFINE
*
LIMIT 3

```

## BACKOUT TRANSACTION

---

```
INPUT 'DEPARTMENT TO BE UPDATED:' #DEPT
IF #DEPT = ' '
  STOP
END-IF
*
FIND EMPLOY-VIEW WITH DEPT = #DEPT
  IF NO RECORDS FOUND
    REINPUT 'NO RECORDS FOUND'
  END-NOREC
  INPUT 'NAME:          ' NAME (AD=0) /
        'LEAVE DUE:    ' LEAVE-DUE (AD=M) /
        'LEAVE TAKEN:' LEAVE-TAKEN (AD=M)
  UPDATE
END-FIND
*
INPUT 'UPDATE TO BE PERFORMED? YES/NO:' #RESP
DECIDE ON FIRST #RESP
  VALUE 'YES'
  END TRANSACTION
  VALUE 'NO'
  BACKOUT TRANSACTION
  NONE
  REINPUT 'PLEASE ENTER YES OR NO'
END-DECIDE
*
END
```

プログラム **BOTEX1** の出力：

```
DEPARTMENT TO BE UPDATED: MGMT30
```

MGMT30 部門の結果：

```
NAME:          POREE
LEAVE DUE:     45
LEAVE TAKEN:   31
```

確認照会：

```
UPDATE TO BE PERFORMED YES/NO: NO
```

# 14 BEFORE BREAK PROCESSING

---

▪ 機能 .....	70
▪ 制限事項 .....	71
▪ 構文説明 .....	71
▪ 例 .....	71

ストラクチャードモード構文

```
BEFORE [BREAK] [PROCESSING]
statement ...
END-BEFORE
```

レポートイングモード構文

```
[BEFORE [BREAK] [PROCESSING]
{
    statement
    DO statement ...DOEND
}
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

---

BEFORE BREAK PROCESSING ステートメントは、次の動作の前に処理を実行するために、自動ブレイク処理と組み合わせて使用します。

- ブレイクのコントロールフィールドの値がチェックされる前
- [AT BREAK](#) ステートメントで指定されたステートメントが実行される前
- Natural システム関数が評価される前

このステートメントは、ブレイク処理に使用するユーザー定義変数の値を初期化したり、計算を行ったりするためによく使用します（[AT BREAK](#) ステートメントを参照）。

このステートメントは非手続き型なので、プログラム内の位置ではなくイベントによって実行されます。

『プログラミングガイド』の次のセクションも参照してください。

- [コントロールブレイク](#)
- [BEFORE BREAK PROCESSING](#) ステートメント
- [BEFORE BREAK PROCESSING](#) ステートメントの例



## 制限事項

- BEFORE BREAK PROCESSING ステートメントは、次のステートメントのいずれかで始まる処理ループにだけ使用できます。
  - FIND
  - READ
  - HISTOGRAM
  - SORT
  - READ WORK FILE

BEFORE BREAK PROCESSING ステートメントは、処理ループ内であればどこでも指定することができます。常にこのステートメントを含む処理ループに関連付けられます。BEFORE BREAK PROCESSING ステートメントは、1つの処理ループ内で1回だけ指定できます。

- BEFORE BREAK PROCESSING ステートメントを、PERFORM BREAK PROCESSING ステートメントと一緒に使用することはできません。

## 構文説明

<i>statement...</i>	ステートメントの例については、下記の「例」を参照してください。  ブレイク処理が行われない（処理ループに <b>AT BREAK</b> ステートメントが指定されていない）場合、BEFORE BREAK PROCESSING ステートメントで指定したステートメントは実行されません。
<b>END-BEFORE</b>	ストラクチャードモードの場合： Natural 予約語 END-BEFORE を使用して、BEFORE BREAK PROCESSING ステートメントを終了させる必要があります。

## 例

```
** Example 'BBPEX1': BEFORE BREAK PROCESSING
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 NAME
  2 SALARY (1)
  2 BONUS (1,1)
*
```

## BEFORE BREAK PROCESSING

---

```
1 #INCOME (P11)
END-DEFINE
*
LIMIT 7
READ EMPLOY-VIEW BY CITY = 'L'
/*
  BEFORE BREAK PROCESSING
    COMPUTE #INCOME = SALARY (1) + BONUS (1,1)
  END-BEFORE
/*
  AT BREAK OF CITY
    WRITE NOTITLE 'AVERAGE INCOME FOR' OLD (CITY) 20X AVER(#INCOME) /
  END-BREAK
/*
  DISPLAY CITY 'NAME' NAME 'SALARY' SALARY (1) 'BONUS' BONUS (1,1)
END-READ
END
```

プログラム **BBPEX1** の出力：

CITY	NAME	SALARY	BONUS	
LA BASSEE	HULOT	165000	70000	
AVERAGE INCOME FOR LA BASSEE				235000
LA CHAPELLE ST LUC	GUILLARD	124100	23000	
LA CHAPELLE ST LUC	BERGE	198500	50000	
LA CHAPELLE ST LUC	POLETTE	124090	23000	
LA CHAPELLE ST LUC	DELAUNEY	115000	23000	
LA CHAPELLE ST LUC	SCHECK	125600	23000	
LA CHAPELLE ST LUC	KREEBS	184550	50000	
AVERAGE INCOME FOR LA CHAPELLE ST LUC				177306

# 15 CALL

---

■ 機能 .....	74
■ 構文説明 .....	74
■ リターンコード .....	75
■ レジスタの使用方法 .....	75
■ 境界の配置 .....	76
■ Adabas コール .....	77
■ ダイレクトロードとダイナミックロード .....	77
■ リンケージの規則 .....	79
■ PL/I プログラムのコール .....	81
■ C プログラムのコール .....	84
■ INTERFACE4 .....	87

CALL [INTERFACE4] *operand1* [USING] [*operand2*] ... 128

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CALL FILE](#) | [CALL LOOP](#) | [CALLNAT](#) | [DEFINE SUBROUTINE](#) | [ESCAPE](#) | [FETCH](#) | [PERFORM](#)

関連機能グループ：「[プログラムおよびルーチンの呼び出し](#)」

## 機能

CALL ステートメントは、Natural プログラムから他の標準プログラミング言語で書かれた外部プログラムを呼び出し、CALL の次のステートメントに戻します。

呼び出される側のプログラムは、標準 CALL インターフェイスをサポートするプログラミング言語で書くことができます。1つ以上の外部プログラムを呼び出すために、複数の CALL ステートメントを指定できます。

TP モニタが CALL インターフェイスをサポートしている場合は、TP モニタの制御下で実行されるプログラム内で CALL ステートメントを発行することもできます。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A	可	不可
<i>operand2</i>	C S A G	A U N P I F B D T L C G	可	可

構文要素の説明：

<b>INTERFACE4</b>	オプションのキーワード INTERFACE4 は、外部プログラムのコールに使用するインターフェイスのタイプを指定します。下記の <b>INTERFACE4</b> を参照してください。
<i>operand1</i>	プログラム名：  呼び出されるプログラムの名前 ( <i>operand1</i> ) は、定数または変数で指定できます。プログラムロジックによって呼び出すプログラムが異なる場合は、長さ 1~8 の英数字の変数として指定します。変数では、プログラム名を左詰めで指定する必要があります。
<b>[USING]</b> <i>operand2</i>	パラメータ：

CALL ステートメントでは、パラメータ (*operand2*) を 128 個まで指定できます。ただし、**INTERFACE4** オプションを使用している場合は、指定できる数が変わります。この場合、パラメータの数は、カタログ化オブジェクトのサイズによって制限されます。Natural オブジェクトにある他のすべてのステートメントによって影響を受けますが、最大 16370 個までのパラメータを使用できます。標準のリンケージ登録規則を使用します。指定する各パラメータフィールドごとに、1つのアドレスがパラメータリストで渡されます。

グループ名を使用した場合、そのグループは個々のフィールドに変換されます。つまり、ユーザーがグループの開始アドレスを指定したい場合、グループの最初のフィールドを指定する必要があります。

**注意:** パック 10 進数の正記号の内部表現は、制御が外部プログラムに渡される前に、NTCMP0 マクロ (コンパイルオプション) の PSIGNF パラメータで指定した値に変更されます。

## リターンコード

呼び出されたプログラムのコンディションコード (Natural に戻ったときのレジスタ 15 の内容) は、Natural システム関数 RET (リターンコード関数) で取得できます。

例:

```
...
RESET #RETURN(B4)
CALL 'PROG1'
IF RET ('PROG1') > #RETURN
  WRITE 'ERROR OCCURRED IN PROGRAM1'
END-IF
...
```

## レジスタの使用方法

レジスタ	内容
R1	パラメータアドレスリストへのアドレスポインタ。
R2	フィールド定義リストへのアドレスポインタ。フィールド定義リストには、パラメータリストで渡される最初の 128 個のフィールドに関する情報が含まれます。各定義は、次の情報を含む 4 バイトのエントリです。 <ul style="list-style-type: none"> <li>■ 1 バイト目には、変数のタイプ (A、B、...) が格納されます。</li> <li>■ タイプ A の変数のサイズが 32767 バイトを超えた場合、変数はタイプ Y として渡されます。</li> <li>■ タイプ B の変数のサイズが 32767 バイトを超えた場合、変数はタイプ X として渡されます。タイプ X と Y は、標準 CALL インターフェイスでロング英数字およびバイナリ変数をサポートするために導入されました。</li> </ul>

レジスタ	内容
	<p>フィールドタイプが N または P の場合：</p> <ul style="list-style-type: none"> <li>■ 2 バイト目には、全体の桁数が格納されます。</li> <li>■ 3 バイト目には、小数点以上の桁数が格納されます。</li> <li>■ 4 バイト目には、小数点以上の桁数が格納されます。</li> </ul> <p>フィールドタイプが X または Y の場合：</p> <ul style="list-style-type: none"> <li>■ 2 バイト目は未使用。</li> <li>■ 3~4 バイト目には、ゼロが格納されます。</li> <li>■ フィールドの長さは R4 を介して渡されます。</li> </ul> <p>その他のフィールドタイプの場合：</p> <ul style="list-style-type: none"> <li>■ 2 バイト目は未使用。</li> <li>■ 3~4 バイト目には、フィールドの長さが格納されます。</li> </ul>
R3	フィールド長のリストへのアドレスポインタ。各長さフィールドは 4 バイトのエントリで、パラメータリストで渡される各フィールドの長さが含まれます。配列の場合、長さは、個々のオカレンスが持つ長さの合計になります。
R4	<p>タイプ X および Y 専用：</p> <ul style="list-style-type: none"> <li>■ サイズが 32767 バイトを超えたタイプ A または B の各変数に対応する、長さが 4 バイトのエントリ。</li> </ul>
R13	18 ワードのセーブエリアのアドレス。
R14	リターンアドレス。
R15	エントリアドレスまたはリターンコード。

## 境界の配置

すべてのユーザー定義変数が保存される Natural データエリアは、常にダブルワード境界から始まります。

**DEFINE DATA** を使用している場合、すべてのデータブロック（例えば、LOCAL、GLOBAL ブロック）はダブルワードで配置され、すべての構造（レベル 1）のフルワードで配置されます。

データエリア内での配置はユーザーの責任で行われ、変数が Natural に定義された順番によって決まります。

## Adabas コール

呼び出される側のプログラムには、Adabasへのコールが含まれていることがあります。呼び出される側のプログラムでは、Adabasオープンまたはクローズコマンドは発行できません。Adabasによって、参照されるデータベースフィールドがすべて開かれます。

Adabas 排他 (EXU) 更新モードを使用する場合は、参照されるすべてのファイル開くために、Natural プロファイルパラメータ OPRB (データベースのオープンとクローズ処理) を使用する必要があります。EXU 更新モードを使用する前に、Natural 管理者に相談してください。

呼び出されたプログラムによってトランザクションを開始または終了する Adabas コマンドが発行された場合、Natural 側ではトランザクションステータスの変化を認識できません。

Adabas へのコールは、それぞれの TP モニタまたはオペレーティングシステム用の Adabas アプリケーションプログラミングインターフェイス (API) の呼び出し規則に適合している必要があります。これは、Natural がサーバーとして動作している場合にも適用されます (例: z/OS または SMARTS の環境下)。

## ダイレクトロードとダイナミックロード

呼び出されたプログラムは Natural ニュークリアスに直接リンクされるか、またはプログラムが最初に呼び出されたときにダイナミックにロードされます。ニュークリアスに直接リンクされる場合、プログラムは、『オペレーション』ドキュメントの「Natural パラメータモジュール」に記載されているプロファイルパラメータ CSTATIC (Natural にスタティックにリンクされるプログラム) で指定されます。

プログラムがダイナミックにロードされる場合は、呼び出されるプログラムを含むロードモジュールライブラリを、Natural 実行 JCL または適切な TP モニタプログラムライブラリにある Natural ロードライブラリに連結する必要があります。詳細については、Natural 管理者に確認してください。

例:

次の例は、国番号を対応する国名に変換する目的で COBOL プログラム TABSUB を呼び出す Natural プログラムを示しています。2つのパラメータフィールドが、Natural プログラムによって TABSUB に渡されます。

- 最初のパラメータは、データベースから読み込まれる国番号です。
- 2番目のパラメータは、対応する国名を返すために使用されます。

## 呼び出す側の Natural プログラム :

```

** Example 'CALEX1': CALL PROGRAM 'TABSUB'
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 BIRTH
  2 COUNTRY
*
1 #COUNTRY      (A3)
1 #COUNTRY-NAME (A15)
1 #FIND-FROM    (D)
1 #FIND-TO      (D)
END-DEFINE
*
MOVE EDITED '19550701' TO #FIND-FROM (EM=YYYYMMDD)
MOVE EDITED '19550731' TO #FIND-TO   (EM=YYYYMMDD)
*
FIND EMPLOY-VIEW WITH BIRTH = #FIND-FROM THRU #FIND-TO
  MOVE COUNTRY TO #COUNTRY
  /*
  CALL 'TABSUB' #COUNTRY #COUNTRY-NAME
  /*
  DISPLAY NAME BIRTH (EM=YYYY-MM-DD) #COUNTRY-NAME
END-FIND
END

```

## 呼び出される側の COBOL プログラム TABSUB :

```

IDENTIFICATION DIVISION.
PROGRAM-ID. TABSUB.
REMARKS. THIS PROGRAM PROVIDES THE COUNTRY NAME
        FOR A GIVEN COUNTRY CODE.
ENVIRONMENT DIVISION.
DATA DIVISION.
WORKING-STORAGE SECTION.
LINKAGE SECTION.
01 COUNTRY-CODE PIC X(3).
01 COUNTRY-NAME PIC X(15).
PROCEDURE DIVISION USING COUNTRY-CODE COUNTRY-NAME.
P-CONVERT.
  MOVE SPACES TO COUNTRY-NAME.
  IF COUNTRY-CODE = 'BLG' MOVE 'BELGIUM' TO COUNTRY-NAME.
  IF COUNTRY-CODE = 'DEN' MOVE 'DENMARK' TO COUNTRY-NAME.
  IF COUNTRY-CODE = 'FRA' MOVE 'FRANCE' TO COUNTRY-NAME.
  IF COUNTRY-CODE = 'GER' MOVE 'GERMANY' TO COUNTRY-NAME.
  IF COUNTRY-CODE = 'HOL' MOVE 'HOLLAND' TO COUNTRY-NAME.
  IF COUNTRY-CODE = 'ITA' MOVE 'ITALY' TO COUNTRY-NAME.
  IF COUNTRY-CODE = 'SPA' MOVE 'SPAIN' TO COUNTRY-NAME.
  IF COUNTRY-CODE = 'UK' MOVE 'UNITED KINGDOM' TO COUNTRY-NAME.

```



P-RETURN.  
GOBACK.

## リンケージの規則

標準のリンケージ登録表記は、バッチモードで使用されます。各TPモニタには、独自の規則があります。これらの規則には従う必要があり、従わなかった場合には予期できない結果が生じることがあります。

次のセクションでは、サポートされている TP モニタに適用される規則について説明します。

- Com-plete を使用した CALL
- CICS を使用した CALL

### Com-plete を使用した CALL

呼び出される側のプログラムは、Com-plete オンラインロードライブラリ内に存在している必要があります。これにより、Com-plete でプログラムをダイナミックにロードすることが可能になります。Com-plete ユーティリティ ULIB を使用すると、プログラムをカタログすることができます。

### CICS を使用した CALL

呼び出される側のプログラムは、CICS ライブラリに連結されたロードモジュールライブラリまたは DFHRPL ライブラリのいずれかに存在している必要があります。また、CICS がプログラムを特定してロードできるように、プログラムは動作している PPT の PPT エントリを保持する必要があります。

リンケージ規則では、パラメータリストアドレスとそれに続くフィールド定義リストアドレスを TWA と COMMAREA の先頭にあるフルワードで渡します。

NCIPARM パラメータモジュール内にあるパラメータ FLDLEN では、フィールド長リストも渡すかどうかを制御します（デフォルトでは、渡しません）。COMMAREA の長さ（0、8、12、または16）は、渡されるリストアドレスの数（0、2、3、または4）を反映しています。最後のアドレスリストは、セットされている上位ビットによって示されます。ユーザーは、TWA または COMMAREA のそれぞれについて、アドレス可能であることを保証する必要があります。これは、ユーザープログラムが Natural に対してスタティックまたはダイレクトリンクプログラムとして定義されていない場合にのみ必要になります。その場合、パラメータリストへのポインタはレジスタ1を介して渡され、さらに定義リストへのポインタはレジスタ2を介して、フィールド長リストへのポインタはレジスタ3を介して、ラージフィールド長リストはレジスタ4を介して渡されます。

NCIPARM パラメータモジュール内にあるパラメータ FLDLEN および COMACAL では、COMMAREA の内容を制御します。

パラメータ値のアドレスリストのアドレスではなく、パラメータ値自体を COMMAREA で渡す必要がある場合は、コールの前に Natural (コールオプション) 端末コマンド %P=C を発行します。

通常、Natural プログラムから Natural 以外のプログラムを呼び出すと、呼び出されたプログラムは会話型の端末 I/O を発行し、Natural スレッドはユーザーがデータを入力するまでブロックされます。Natural スレッドがブロックされるのを防ぐには、端末コマンド %P=V を使用できます。

通常、Natural プログラムで CICS 環境下の Natural 以外のプログラムを呼び出す場合、コールは EXEC CICS LINK 要求によって実行されます。コールに標準リンクを使用する場合は、端末コマンド %P=S を発行します。(この場合、呼び出されたされたプログラムでは、標準的な方法でレジスタを使用することで、標準のリンケージ規則に従う必要があります)。

31 ビットモード環境では、次の処理が適用されます。AMODE=24 にリンクされたプログラムを呼び出し、またスレッドが 16 MB よりも上にある場合は、「値によるコール」が自動的に実行されます。つまり、呼び出されたプログラムに渡すように指定されたパラメータは、16MB境界よりも下にコピーされます。

### CICS 環境下でのリターンコード

CICS 自体は CICS の規則に従ったコール (EXEC CICS LINK) の戻り値はサポートしていませんが、C/C++ プログラムのコールでは、例外として、exit() 関数または return() ステートメントが渡す値は EIBRESP2 フィールドに保存されます。ただし、Natural CICS インターフェイスでは CALL ステートメントのリターンコードがサポートされているため、呼び出されたプログラムから制御が戻ると、Natural によって最初に EIBRESP2 フィールドにゼロ以外のリターンコードがあるかどうかチェックされます。次に、COMMAREA の最初のフルワードが変更されたかどうかチェックされます。変更されている場合は、その新しい内容がリターンコードになります。変更されていない場合は、TWA がチェックされ、その新しい内容がリターンコードになります。2つのフルワードがいずれも変更されていない場合、リターンコードは 0 になります。



**注意:** パラメータ値を COMMAREA で渡した場合 (%P=C)、EIBRESP2 フィールドのリターンコードだけがチェックされます。つまり、C/C++ 以外のプログラムでは、リターンコードは常に 0 になります。

### CICS の使用例:

```
IDENTIFICATION DIVISION.
PROGRAM-ID. TABSUB.
REMARKS. THIS PROGRAM PERFORMS A TABLE LOOK-UP AND
          RETURNS A TEXT MESSAGE.
ENVIRONMENT DIVISION.
DATA DIVISION.
WORKING-STORAGE SECTION.
01 MSG-TABLE.
   03 FILLER          PIC X(15) VALUE 'MESSAGE1      '.
   03 FILLER          PIC X(15) VALUE 'MESSAGE2      '.
```

```
03 FILLER      PIC X(15) VALUE 'MESSAGE3  ' .
03 FILLER      PIC X(15) VALUE 'MESSAGE4  ' .
03 FILLER      PIC X(15) VALUE 'MESSAGE5  ' .
03 FILLER      PIC X(15) VALUE 'MESSAGE6  ' .
03 FILLER      PIC X(15) VALUE 'MESSAGE7  ' .
01 TAB REDEFINES MSG-TABLE.
    03 MESSAGE OCCURS 7 TIMES PIC X(15).
LINKAGE SECTION.
01 TWA-DATA.
    03 PARM-POINTER USAGE IS POINTER.
01 PARM-LIST.
    03 DATA-LOC-IN  USAGE IS POINTER.
    03 DATA-LOC-OUT USAGE IS POINTER.
01 INPUT-DATA.
    03 INPUT-NUMBER      PIC 99.
01 OUTPUT-DATA.
    03 OUTPUT-MESSAGE    PIC X(15).
PROCEDURE DIVISION.
100-INIT.
    EXEC CICS ADDRESS TWA(ADDRESS OF TWA-DATA) END-EXEC.
    SET ADDRESS OF PARM-LIST  TO PARM-POINTER.
    SET ADDRESS OF INPUT-DATA TO DATA-LOCIN.
    SET ADDRESS OF OUTPUT-DATA TO DATA-LOC-OUT.
200-PROCESS.
    MOVE MESSAGE (INPUT-NUMBER) TO OUTPUT-MESSAGE.
300-RETURN.
    EXEC CICS RETURN END-EXEC.
400-DUMMY.
    GO-BACK.
```

## PL/I プログラムのコール

PL/I で書かれた呼び出される側のプログラムでは、次の追加手順が必要です。

- パラメータリストは標準のリストで、他の PL/I プログラムから渡される引数リストではないため、渡されるアドレスは LOCATOR DESCRIPTOR を指していません。この問題は、パラメータフィールドを算術変数として定義することで解決できます。これにより、PL/I では、パラメータリストが LOCATOR DESCRIPTOR コントロールブロックとしてではなくデータのアドレスとして扱われるようになります。

パラメータフィールドを定義するための推奨される手法について、次の例で説明します。

```
PLIPROG: PROC(INPUT_PARM_1, INPUT_PARM_2) OPTIONS(MAIN);
  DECLARE (INPUT_PARM_1, INPUT_PARM_2) FIXED;
  PTR_PARM_1 = ADDR(INPUT_PARM_1);
  PTR_PARM_2 = ADDR(INPUT_PARM_2);
  DECLARE FIRST_PARM      PIC '99'    BASED (PTR_PARM_1);
  DECLARE SECOND_PARM     CHAR(12)    BASED (PTR_PARM_2);
```

入力リストの各パラメータは、一意な要素として扱われる必要があります。入力パラメータの数は、Naturalプログラムから渡される数と完全に一致している必要があります。入力パラメータとそれらの属性は、Naturalの定義と一致している必要があります。一致していない場合は、予測できない結果が生じることがあります。PL/I でパラメータを引き渡す方法の詳細については、関連のIBM PL/I ドキュメントを参照してください。

以下では次のトピックについて説明します。

- [PL/I プログラムのコールの例](#)
- [CICS 環境下で動作している PL/I プログラムのコールの例](#)

## PL/I プログラムのコールの例

```
** Example 'CALEX2': CALL PROGRAM 'NATPLI'
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 AREA-CODE
  2 REDEFINE AREA-CODE
    3 #AC      (N1)
*
1 #INPUT-NUMBER  (N2)
1 #OUTPUT-COMMENT (A15)
END-DEFINE
*
READ EMPLOY-VIEW IN LOGICAL SEQUENCE BY NAME
      STARTING FROM 'WAGNER'
  MOVE ' ' TO #OUTPUT-COMMENT
  MOVE #AC TO #INPUT-NUMBER
/*
  CALL 'NATPLI' #INPUT-NUMBER #OUTPUT-COMMENT
/*
END-READ
*
END
```

呼び出される側の PL/I プログラム NATPLI :

```
NATPLI: PROC(PARM_COUNT, PARM_COMMENT) OPTIONS(MAIN);
  /*                                     */
  /* THIS PROGRAM ACCEPTS AN INPUT NUMBER */
  /* AND TRANSLATES IT TO AN OUTPUT CHARACTER */
  /* STRING FOR PLACEMENT ON THE FINAL */
  /* NATURAL REPORT */
  /*                                     */
  /*                                     */
  /*                                     */
  DECLARE PARM_COUNT, PARM_COMMENT  FIXED;
  DECLARE ADDR BUILTIN;
  COUNT_PTR = ADDR(PARM_COUNT);
  COMMENT_PTR = ADDR(PARM_COMMENT);
  DECLARE INPUT_NUMBER PIC '99' BASED (COUNT_PTR);
  DECLARE OUTPUT_COMMENT CHAR(15) BASED (COMMENT_PTR);
  DECLARE COMMENT_TABLE(9) CHAR(15) STATIC INITIAL
    ('COMMENT1 ',
     'COMMENT2 ',
     'COMMENT3 ',
     'COMMENT4 ',
     'COMMENT5 ',
     'COMMENT6 ',
     'COMMENT7 ',
     'COMMENT8 ',
     'COMMENT9 ');
  OUTPUT_COMMENT = COMMENT_TABLE(INPUT_NUMBER);
  RETURN;
END NATPLI;
```

### CICS 環境下で動作している PL/I プログラムのコールの例

```
** Example 'CALEX3': CALL PROGRAM 'CICSP'
*****
DEFINE DATA LOCAL
1 #MESSAGE (A10) INIT <' '>
END-DEFINE
*
CALL 'CICSP' #MESSAGE
DISPLAY #MESSAGE
*
END
```

呼び出される側の PL/I プログラム CICSP :

```
CICSP: PROCEDURE OPTIONS (MAIN REENTRANT);
  DCL 1      TWA_ADDRESS    BASED(TWA_POINTER);
      2      LIST_ADDRESS  POINTER;
  DCL 1 PTR_TO_LIST      BASED(LIST_ADDRESS);
      2 PARM_01          POINTER;
  DCL MESSAGE CHAR(10) BASED(PARM_01);
  EXEC CICS ADDRESS TWA(TWA_POINTER);
  MESSAGE='SUCCESS'; EXEC CICS RETURN; END CICSP;
```

## C プログラムのコール

C プログラムを使用する前に、そのプログラムをコンパイルしてリンクする必要があります。

- 例えば、IBM の C コンパイラを使用して実行可能モジュールをビルドする場合。IBM の C コンパイラでは LE コードが生成されるため、サンプルは LE 環境でなければ実行できません。LE プログラムを実行するには、Natural フロントエンドを LE に対応できるようにインストールする必要があります。
- Dignus や SASC などの他の C コンパイラを使用する場合は、C 以外の環境からコールできるモジュールをビルドする必要があります。詳細については、該当するコンパイラのドキュメントを参照してください。
- C プログラムには、インクルードファイル NATUSER をインクルードする必要があります。

INTERFACE4 用に使われた C プログラムは、メインフレームシステムだけでなく、UNIX、OpenVMS、または Windows システムでも使用できます。一方、標準 CALL インターフェイス用に使われた C プログラムは、プラットフォームに依存します。

CALL INTERFACE4 によって C プログラムを呼び出す場合、または Natural サブプログラムが C プログラムから呼び出される場合は、NATXCAL4 を実行可能モジュールにリンクする必要があります。INTERFACE4 コールバック関数の 1 つを使用して、パラメータ定義とパラメータ値を取得します。コールバック関数については、次で説明します。

関数 `ncxr_if4_callnat` を使用して、C プログラムから Natural サブプログラムを実行します。

プロトタイプ：

```
int ncxr_if4_callnat ( char *natpgm, int parmnum, struct parameter_description
*descr );
```

パラメータの説明：

<b>natpgm</b>	呼び出される Natural サブプログラムの名前。	
<b>parmnum</b>	サブプログラムに渡されるパラメータフィールドの数。	
<b>descr</b>	struct parameter_description のアドレス。この構造の詳細については、「 <a href="#">Interface4 のオペランド構造</a> 」を参照してください。	
<b>return</b>	戻り値：	情報：
	0	OK  サブプログラムの実行中に Natural エラーが発生した場合、エラーに関する情報が値 natpgm によって *NAT nnnn の形式で返されます。nnnn は Natural エラー番号に対応しています。
	-1	無効なパラメータ番号。
	-2	内部エラーが発生しました。

以下では次のトピックについて説明します。

- [標準 CALL による C プログラムのコールの例](#)
- [CALL INTERFACE4 による C プログラムのコールの例](#)

### 標準 CALL による C プログラムのコールの例

```

** Example 'CALEX4': CALL PROGRAM 'ADD'
*****
DEFINE DATA LOCAL
1 #OP1 (I4)
1 #OP2 (I4)
1 #SUM (I4)
END-DEFINE
*
CALL 'ADD' #OP1 #OP2 #SUM
DISPLAY #SUM
*
END

```

呼び出される側の C プログラム ADD :

```
/*
** Example C Program ADD.c
*/
NATFCT ADD (int *op1, int *op2, int *sum)
{
*sum = *op1 + *op2;    /* add opperands */

return 0;              /* return successfully */
} /* ADD */
```

### CALL INTERFACE4 による C プログラムのコールの例

```
** Example 'CALEX5': CALL PROGRAM 'ADD4'
*****
DEFINE DATA LOCAL
1 #OP1  (I4)
1 #OP2  (I4)
1 #SUM  (I4)
END-DEFINE
*
CALL INTERFACE4 'ADD4' #OP1 #OP2 #SUM
DISPLAY #SUM
*
END
```

呼び出される側の C プログラム ADD4 :

```
NATFCT ADD4 NATARGDEF(numparm, parmhandle, parmdec)
{
NATTYP_I4 op1, op2, sum;          /* local integers */
int i;                            /* loop counter */
struct parameter_description desc;
int rc;                            /* return code access functions */

/*
** test number of arguments
*/
if (numparm != 3) return 1;

/*
** test types of arguments
*/
for (i = 0; i < (int) numparm; i++)
{
    rc = ncxr_get_parm_info( i, parmhandle, &desc );
    if ( rc ) return rc;

    if ( desc.format != 'I' || desc.length != sizeof(NATTYP_I4) || desc.dimensions
```



```
!= 0 )
    {
        return 2;          /* invalid parameter */
    }
}

/*
** get arguments
*/
rc = ncxr_get_parm( 0, parmhandle, sizeof op1, (void *)&op1 );
if ( rc ) return rc;

rc = ncxr_get_parm( 1, parmhandle, sizeof op2, (void *)&op2 );
if ( rc ) return rc;

/*
** perform the addition
*/
sum = op1 + op2;

/*
** move the result back to operand 3
*/
rc = ncxr_put_parm( 2, parmhandle, sizeof sum, (void *)&sum );
if ( rc ) return rc;

/*
** all ok, return success to the caller
*/
return 0;
} /* ADD4 */
```

## INTERFACE4

INTERFACE4 キーワードでは、外部プログラムのコールに使用するインターフェイスのタイプを指定します。このキーワードはオプションです。このキーワードを指定する場合は、インターフェイスを **Interface4** として定義し、外部プログラムのコールに使用します。

次の表に、INTERFACE4 を使用した場合と、INTERFACE4 使用しなかった場合の CALL ステートメントの違いを示します。

	キーワードINTERFACE4を使用しないCALLステートメント	キーワードINTERFACE4を使用するCALLステートメント
使用可能なパラメータ数	128	16370 個以下
1つのパラメータの最大データサイズ	制限なし	1 GB
配列情報の取得	無	有
ラージオペランドとダイナミックオペランドのサポート	完全な読み取りアクセス、オペランドのサイズを変更しない書き込み	有
API 経由でのパラメータアクセス	ダイレクト	API 経由

パラメータの最大数は、生成プログラム (GP) の最大サイズおよびステートメントの最大サイズによって制限されます。プログラムに CALL ステートメントのみが含まれている場合は、16370 個のパラメータが使用できます。他のステートメントを使用すると、最大数は減少します。

以下では次のトピックについて説明します。

- [INTERFACE4 - 外部 3GL プログラムインターフェイス](#)
- [INTERFACE4 のオペランド構造体](#)
- [INTERFACE4 - パラメータアクセス](#)
- [エクスポート機能](#)

## INTERFACE4 - 外部 3GL プログラムインターフェイス

外部 3GL プログラムのインターフェイスは、Natural の CALL ステートメントで INTERFACE4 を指定するときに、次のように定義します。

```
NATFCT functionname (numparm, parmhandle, traditional)
```

USR_WORD	numparm;	転送するオペランド ( <i>operand2</i> ) の総数が格納される 16 ビット unsigned short 型整数。
void	*parmhandle;	構造体を渡すパラメータへのポインタ。
void	*traditional;	インターフェイスタイプのチェックに使用します (NULL ポインタでない場合は通常の CALL インターフェイス)。

## INTERFACE4 のオペランド構造体

INTERFACE4 のオペランド構造体の名前は、「parameter\_description」で次のように定義します。構造体は、ヘッダーファイル *natuser.h* とともに提供されます。

struct parameter_description		
void *	address	パラメータデータのアドレス。不整列。realloc() および free() は使用できません。
int	format	フィールドデータフォーマット：NCXR_TYPE_ALPHA など ( <i>natuser.h</i> )。
int	length	長さ (小数点以上、該当する場合)。
int	precision	小数点以下の長さ (該当する場合)。
int	byte_length	dimensions の整数値のバイト単位のフィールドの長さ (0～IF4_MAX_DIM)。
int	dimensions	次元の数 (0～IF4_MAX_DIM)。
int	length_all	配列のデータ長の合計 (バイト単位)。
int	flags	ビットごとに OR で結合したフラグビットは次のとおりです。 IF4_FLG_PROTECTED：           パラメータは書き込み保護されています。 IF4_FLG_DYNAMIC：            パラメータはダイナミック変数です。 IF4_FLG_NOT_CONTIGUOUS： 配列の要素が隣接していません (要素間に空白があります)。 IF4_FLG_AIV：                このパラメータはアプリケーション独立変数です。 IF4_FLG_DYNVAR：            パラメータはダイナミック変数です。 IF4_FLG_XARRAY：            パラメータは x-array です。 IF4_FLG_LBVAR_0：           次元 0 の下限は可変です。 IF4_FLG_UBVAR_0：           次元 0 の上限は可変です。 IF4_FLG_LBVAR_1：           次元 1 の下限は可変です。 IF4_FLG_UBVAR_1：           次元 1 の上限は可変です。 IF4_FLG_LBVAR_2：           次元 2 の下限は可変です。 IF4_FLG_UBVAR_2：           次元 2 の上限は可変です。
int	occurrences[IF4_MAX_DIM]	各次元における配列のオカレンス。
int	indexfactors[IF4_MAX_DIM]	各次元の配列インデックスファクタ。
void *	dynp	内部使用に予約済み。
void *	pops	内部使用に予約済み。

アドレス要素は、ダイナミック変数の配列および x-array に対してヌルです。この場合、配列データは全体としてアクセスすることはできませんが、下記のパラメータアクセス機能でアクセスする必要があります。

固定長の変数の固定範囲の配列に対して、配列の内容はアドレス要素を使用してダイレクトにアクセスすることができます。この場合、配列要素 (i,j,k) のアドレスは以下のように計算されます (特に配列要素が連続していない場合)。

```
elementaddress = address + i * indexfactors[0] + j * indexfactors[1] + k *  
indexfactors[2]
```

配列の次元数が3次元よりも小さい場合は、最後の項が除外されます。

## INTERFACE4・パラメータアクセス

パラメータのアクセスに使用する機能の集合があります。処理フローは次のとおりです。

- 3GL プログラムは INTERFACE4 オプション付きの CALL ステートメントで呼び出され、パラメータが 3GL プログラムに渡されます (上記を参照)。
- その後、3GL プログラムではパラメータデータ自体またはフォーマット、長さ、配列情報などのパラメータに関する情報を検索するために、Natural のエクスポート機能を使用することができます。
- **エクスポート機能**はパラメータデータを戻すことにも使用できます。

3GL プログラムから任意のサブプログラムをコールするために、新規パラメータセットを作成し、初期化する機能もあります。この機能を使用して、3GL プログラムがメモリを上書きしないようにするために、パラメータのアクセスが保証されます (Natural データは安全で、メモリは 3GL プログラムのデータ内を上書きすることはまだ可能です)。

## エクスポート機能

以下では次のトピックについて説明します。

- **パラメータ情報の取得**
- **パラメータデータの取得**
- **オペランドデータの書き戻し**
- **パラメータセットの作成、初期化、削除**
- **パラメータセットの作成**
- **パラメータセットの削除**
- **スタティックデータタイプのスカラの初期化**
- **スタティックデータタイプの配列の初期化**
- **ダイナミックデータタイプのスカラの初期化**
- **ダイナミックデータタイプの配列の初期化**

## ■ X-array パラメータのサイズ変更

### パラメータ情報の取得

この機能は、任意のパラメータから必要な情報をすべて受け取るために、3GL プログラムで使用します。情報は、struct parameter\_description (上記を参照) に返されます。

プロトタイプ：

```
int ncxr_get_parm_info ( int parmnum, void *parmhandle, struct parameter_description *descr );
```

パラメータの説明：

<b>parmnum</b>	パラメータの順序を表す番号。これにより、渡されたパラメータリストのパラメータが識別されます。範囲：0～numparm-1。	
<b>parmhandle</b>	内部パラメータ構造へのポインタ	
<b>descr</b>	struct parameter_description のアドレス	
<b>return</b>	戻り値：	情報：
	0	OK
	-1	無効なパラメータ番号。
	-2	内部エラーが発生しました。
	-7	インターフェイスバージョンの競合。

### パラメータデータの取得

この機能は、任意のパラメータからデータを取得するために、3GL プログラムで使用します。

Natural は、指定したパラメータ番号でパラメータを識別し、指定したバッファサイズのアドレスにパラメータデータを書き出します。

パラメータデータが指定したバッファサイズより長い場合、Natural は指定した長さにそのデータを切り捨てます。外部 3GL プログラムは ncxr\_get\_parm\_info 関数を、パラメータデータの長さを要求するために使用することができます。

パラメータデータを取得する関数は 2 つあり、ncxr\_get\_parm はパラメータ全体を取得し (パラメータが配列の場合でも)、ncxr\_get\_parm\_array 関数は特定の配列要素を取得します。

(ダイナミックまたはスタティックな) 3GL プログラムにより、指定したサイズのメモリが「バッファ」に対して割り当てられない場合、予期できない結果になることがあります。Natural では NULL ポインタのチェックだけが実行されます。

タイプ I2/I4/F4/F8 の変数で、データが切り取られる場合 (バッファ長がパラメータの合計長と等しくない)、その結果はマシンのタイプにより異なります (リトルエンディアン/ビッグ

エンディアン)。アプリケーションによっては、反復を可能にするために、スタティックデータを使用しないようにユーザー出口をプログラムする必要があります。

プロトタイプ：

```
int ncxr_get_parm( int parmnum, void *parmhandle, int buffer_length, void *buffer
) int ncxr_get_parm_array( int parmnum, void *parmhandle, int buffer_length, void
*buffer, int *indexes )
```

この機能は、各次元のインデックスを指定可能であることを除けば、ncxr\_get\_parmと同じです。未使用の次元のインデックスは0と指定しておく必要があります。

パラメータの説明：

<b>parmnum</b>	パラメータの順序を表す番号。これにより、渡されたパラメータリストのパラメータが識別されます。範囲：0～numparm-1。	
<b>parmhandle</b>	内部パラメータ構造へのポインタ	
<b>buffer_length</b>	要求したデータを書き出す必要のあるバッファの長さ。	
<b>buffer</b>	要求したデータを書き出す必要のあるバッファのアドレス。このバッファは I2/I4/F4/F8 変数にアクセスしやすいように配置しておく必要があります。	
<b>indexes</b>	インデックス情報をもつ配列	
<b>return</b>	戻り値：	情報：
	< 0	情報検索時のエラーを示します。
	-1	無効なパラメータ番号。
	-2	内部エラーが発生しました。
	-3	データは切り捨てられました。
	-4	データが配列ではありません。
	-7	インターフェイスバージョンの競合。
	-100	次元 0 のインデックスが範囲外です。
	-101	次元 1 のインデックスが範囲外です。
	-102	次元 2 のインデックスが範囲外です。
	0	操作に成功しました。
	> 0	正常に行われましたが、（バッファの方がデータよりも長かったため）データの長さはこのバイト数になりました。

## オペランドデータの書き戻し

この機能は、任意のパラメータヘデータを戻すために、3GLプログラムで使用します。Naturalでは、指定したパラメータ番号でパラメータを識別し、指定したバッファサイズとアドレスからパラメータデータを書き出します。パラメータデータが指定したバッファサイズより小さい場合、データはパラメータデータの長さに切り捨てられます。つまり、バッファの残りは無視されます。パラメータデータが指定したバッファサイズより大きい場合、データは指定したバッファ長だけコピーされます。パラメータの残りはそのままです。これは、配列にも同様に当てはまります。パラメータとしてのダイナミック変数については、パラメータのサイズが指定したバッファ長に変更されます。

タイプ I2/I4/F4/F8 の変数で、データが切り取られる場合（バッファ長がパラメータの合計長と等しくない）、その結果はマシンのタイプにより異なります（リトルエンディアン/ビッグエンディアン）。アプリケーションによっては、反復を可能にするために、スタティックデータを使用しないようにユーザー出口をプログラムする必要があります。

プロトタイプ：

```
int ncxr_put_parm ( int parmnum, void *parmhandle, int buffer_length, void *buffer );
int ncxr_put_parm_array ( int parmnum, void *parmhandle, int buffer_length, void *buffer, int *indexes );
```

パラメータの説明：

<b>parmnum</b>	パラメータの順序を表す番号。これにより、渡されたパラメータリストのパラメータが識別されます。範囲：0～numparm-1。	
<b>parmhandle</b>	内部パラメータ構造体を指すポインタ。	
<b>buffer_length</b>	データを送るバッファのバッファアドレスを元に複写するためのデータの長さ。	
<b>indexes</b>	インデックス情報。	
<b>return</b>	戻り値：	情報：
	< 0	情報複写時のエラーを示します。
	-1	無効なパラメータ番号。
	-2	内部エラーが発生しました。
	-3	指定したデータが多すぎます。パラメータの長さで複写されました。
	-4	パラメータが配列ではありません。
	-5	パラメータは保護されています（定数または AD=0）。
	-6	「メモリ不足」により、ダイナミック変数のサイズを変更できませんでした。
	-7	インターフェイスバージョンの競合。
-13	指定したバッファに、不完全な Unicode 文字が含まれています。	

-100	次元 0 のインデックスが範囲外です。
-101	次元 1 のインデックスが範囲外です。
-102	次元 2 のインデックスが範囲外です。
0	操作に成功しました。
> 0	正常に行われましたが、パラメータ長が指定された長さを超えていたため、パラメータの長さはこのバイト数になりました。

## パラメータセットの作成、初期化、削除

3GL プログラムが Natural サブプログラムを呼び出すには、サブプログラムが予期しているパラメータに対応するパラメータセットを構築する必要があります。関数 `ncxr_create_parm` を使用して、`ncxr_if_callnat` へのコールで渡すパラメータセットを作成します。作成されたパラメータセットは、`CALL INTERFACE4` ステートメントで 3GL プログラムに渡されるパラメータセットのように、`opaque` パラメータハンドルによって表されます。したがって、上記のように、新しく作成したパラメータセットは関数 `ncxr_put_parm*` および `ncxr_get_parm*` で操作することができます。

新しく作成したパラメータセットは、関数 `ncxr_create_parm` を呼び出した後は、まだ初期化されていません。個々のパラメータは下記の `ncxr_parm_init*` 関数のセットによって特定データタイプに初期化されます。その後、関数 `ncxr_put_parm*` および `ncxr_get_parm*` は個々のパラメータの内容にアクセスするために使用されます。呼び出す側がパラメータセットを終了した後、それらはパラメータハンドルを削除する必要があります。したがって、`ncxr_if4_callnat` を通して呼び出されるサブプログラムのパラメータセットの作成と使用の一般的なシーケンスは、次のようになります。

```
ncxr_create_parm
ncxr_init_parm*
ncxr_init_parm*
...
ncxr_put_parm*
ncxr_put_parm*
...
ncxr_get_parm_info*
ncxr_get_parm_info*
...
ncxr_if4_callnat
...
ncxr_get_parm_info*
ncxr_get_parm_info*
...
ncxr_get_parm*
ncxr_get_parm*
...
ncxr_delete_parm
```



## パラメータセットの作成

関数 `ncxr_create_parm` を使用して、`ncxr_if_callnat` へのコールで渡すパラメータセットを作成します。

プロトタイプ：

```
int ncxr_create_parm( int parmnum, void** pparmhandle )
```

パラメータの説明：

<b>parmnum</b>	作成するパラメータの数。	
<b>pparmhandle</b>	作成されたパラメータハンドルへのポインタ。	
<b>return</b>	戻り値：	情報：
	< 0	エラー：
	-1	パラメータカウントが不正です。
	-2	内部エラーが発生しました。
	-6	メモリ不足の状態です。
	0	操作に成功しました。

## パラメータセットの削除

`ncxr_create_parm` で作成されたパラメータセットを削除するには、関数 `ncxr_delete_parm` を使用します。

プロトタイプ：

```
int ncxr_delete_parm( void* parmhandle )
```

パラメータの説明：

<b>parmhandle</b>	削除するパラメータハンドルへのポインタ。	
<b>return</b>	戻り値：	情報：
	< 0	エラー：
	-2	内部エラーが発生しました。
	0	操作に成功しました。

## スタティックデータタイプのスカラの初期化

プロトタイプ：

```
int ncxr_init_parm_s( int parmnum, void *parmhandle, char format, int length, int precision, int flags );
```

パラメータの説明：

<b>parmnum</b>	パラメータの順序を表す番号。これにより、渡されたパラメータリストのパラメータが識別されます。範囲：0～numparm-1。	
<b>parmhandle</b>	パラメータハンドルへのポインタ。	
<b>format</b>	パラメータのフォーマット。	
<b>length</b>	パラメータの長さ。	
<b>precision</b>	パラメータの精度。	
<b>flags</b>	フラグの組み合わせ。 IF4_FLG_PROTECTED	
<b>return</b>	戻り値：	情報：
	< 0	エラー：
	-1	無効なパラメータ番号。
	-2	内部エラーが発生しました。
	-6	メモリ不足の状態です。
	-8	フォーマットが正しくありません。
	-9	無効な長さまたは精度。
0	操作に成功しました。	

## スタティックデータタイプの配列の初期化

プロトタイプ：

```
int ncxr_init_parm_sa( int parmnum, void *parmhandle, char format, int length, int precision, int dim, int *occ, int flags );
```

パラメータの説明：

<b>parmnum</b>	パラメータの順序を表す番号。これにより、渡されたパラメータリストのパラメータが識別されます。範囲：0～numparm-1。	
<b>parmhandle</b>	パラメータハンドルへのポインタ。	
<b>format</b>	パラメータのフォーマット。	
<b>length</b>	パラメータの長さ。	
<b>precision</b>	パラメータの精度。	
<b>dim</b>	配列の次元。	
<b>occ</b>	次元ごとのオカレンス数。	
<b>flags</b>	フラグの組み合わせ。  IF4_FLG_PROTECTED IF4_FLG_LBVAR_0 IF4_FLG_UBVAR_0 IF4_FLG_LBVAR_1 IF4_FLG_UBVAR_1 IF4_FLG_LBVAR_2 IF4_FLG_UBVAR_2	
<b>return</b>	戻り値：	情報：
	< 0	エラー：
	-1	無効なパラメータ番号。
	-2	内部エラーが発生しました。
	-6	メモリ不足の状態です。
	-8	フォーマットが正しくありません。
	-9	無効な長さまたは精度。
	-10	無効な次元数。
	-11	変数制限の無効な組み合わせ。
0	操作に成功しました。	

### ダイナミックデータタイプのスカラの初期化

プロトタイプ：

```
int ncxr_init_parm_d( int parmnum, void *parmhandle, char format, int flags );
```

パラメータの説明：

<b>parmnum</b>	パラメータの順序を表す番号。これにより、渡されたパラメータリストのパラメータが識別されます。範囲：0～numparm-1。	
<b>parmhandle</b>	パラメータハンドルへのポインタ。	
<b>format</b>	パラメータのフォーマット。	
<b>flags</b>	フラグの組み合わせ。 IF4_FLG_PROTECTED	
<b>return</b>	戻り値：	情報：
	< 0	エラー：
	-1	無効なパラメータ番号。
	-2	内部エラーが発生しました。
	-6	メモリ不足の状態です。
	-8	フォーマットが正しくありません。
	0	操作に成功しました。

### ダイナミックデータタイプの配列の初期化

プロトタイプ：

```
int ncxr_init_parm_da( int parmnum, void *parmhandle, char format, int dim, int *occ, int flags );
```

パラメータの説明：

<b>parmnum</b>	パラメータの順序を表す番号。これにより、渡されたパラメータリストのパラメータが識別されます。範囲：0～numparm-1。	
<b>parmhandle</b>	パラメータハンドルへのポインタ。	
<b>format</b>	パラメータのフォーマット。	
<b>dim</b>	配列の次元。	
<b>occ</b>	次元ごとのオカレンス数。	
<b>flags</b>	フラグの組み合わせ。 IF4_FLG_PROTECTED IF4_FLG_LBVAR_0 IF4_FLG_UBVAR_0 IF4_FLG_LBVAR_1 IF4_FLG_UBVAR_1 IF4_FLG_LBVAR_2 IF4_FLG_UBVAR_2	
<b>return</b>	戻り値：	情報：
	< 0	エラー：

-1	無効なパラメータ番号。
-2	内部エラーが発生しました。
-6	メモリ不足の状態です。
-8	フォーマットが正しくありません。
-10	無効な次元数。
-11	変数制限の無効な組み合わせ。
0	操作に成功しました。

### X-array パラメータのサイズ変更

プロトタイプ：

```
int ncxr_resize_parm_array( int parmnum, void *parmhandle, int *occ );
```

パラメータの説明：

<b>parmnum</b>	パラメータの順序を表す番号。これにより、渡されたパラメータリストのパラメータが識別されます。範囲：0～numparm-1。	
<b>parmhandle</b>	パラメータハンドルへのポインタ。	
<b>occ</b>	次元当たりの新規オカレンス数。	
<b>return</b>	戻り値：	情報：
	<0	エラー：
	-1	無効なパラメータ番号。
	-2	内部エラーが発生しました。
	-6	メモリ不足の状態です。
	-12	オペランドが（指定された次元の1つで）サイズ変更可能ではありません。
0	操作に成功しました。	

すべての関数プロトタイプがファイル *natuser.h* で宣言されます。

---

# 16 CALL FILE

---

▪ 機能 .....	102
▪ 制限事項 .....	102
▪ 構文説明 .....	103
▪ 例 .....	103

ストラクチャードモード構文

```
CALL FILE 'program-name' operand1 operand2  
statement ...  
END-FILE
```

レポートモード構文

```
CALL FILE 'program-name' operand1 operand2  
statement ...  
[LOOP]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CALL](#) | [CALL LOOP](#) | [CALLNAT](#) | [DEFINE SUBROUTINE](#) | [ESCAPE](#)  
| [FETCH](#) | [PERFORM](#)

関連機能グループ：「[プログラムおよびルーチンの呼び出し](#)」

## 機能

---

CALL FILE ステートメントは、Adabas 以外のファイルからレコードを読み取って Natural プログラムに返す、Natural 以外のプログラムを呼び出すために使用します。

## 制限事項

---

ステートメント [AT BREAK](#)、[AT START OF DATA](#)、および [AT END OF DATA](#) を、CALL FILE 処理ループ内で使用することはできません。



## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A	A U N P I F B D T L C	可	可
<i>operand2</i>	S A G	A U N P I F B D T L C	可	可

構文要素の説明：

' <i>program-name</i> '	呼び出される Natural 以外のプログラムの名前。
<i>operand1</i>	コントロールフィールド： <i>operand1</i> は制御情報の提供に使用されます。
<i>operand2</i>	<i>operand2</i> では、レコードエリアを定義します。 読み取るレコードのフォーマットは、レコード内の最初のフィールド名に続けてフィールド定義（または FILLER <i>nX</i> ）エントリを使用して定義します。レコードフォーマットを定義するために使用されるフィールドは、Natural プログラム内で事前に定義しないでください。これは Natural がフィールドを連続ストレージに割り付けることを保証するためです。
<i>statement ...</i>	CALL FILE ステートメントは、処理ループを開始します。この処理ループは、ESCAPE または STOP ステートメントで終了する必要があります。さまざまな条件に対応して CALL FILE ループから抜け出せるように、複数の ESCAPE ステートメントを指定することも可能です。
END-FILE	END-FILE ステートメントを使用して、処理ループを閉じる必要があります。

## 例

呼び出す側のプログラム：

```

** Example 'CFIEX1': CALL FILE
*****
DEFINE DATA LOCAL
1 #CONTROL (A3)
1 #RECORD
  2 #A      (A10)
  2 #B      (N3.2)
  2 #FILL1  (A3)
  2 #C      (P3.1)
END-DEFINE
*
```

```
CALL FILE 'USER1' #CONTROL #RECORD
  IF #CONTROL = 'END'
    ESCAPE BOTTOM
  END-IF
END-FILE
/*****
/* ... PROCESS RECORD ...
/*****
END
```

上記の例で呼び出されたプログラムが、Naturalプログラムに渡すレコードのバイトレイアウトは、次のとおりです。

```
CONTROL #A      #B  FILLER #C
(A3)   (A10)   (N3.2) 3X   (P3.1)

xxx xxxxxxxxxxx xxxxx   xxx   xxx
```

呼び出される側の **COBOL** プログラム：

```
ID DIVISION.
PROGRAM-ID. USER1.
ENVIRONMENT DIVISION.
INPUT-OUTPUT SECTION.
FILE-CONTROL.
  SELECT USRFILE ASSIGN UT-S-FILEUSR.
DATA DIVISION.
FILE SECTION.
FD  USRFILE RECORDING F LABEL RECORD OMITTED
  DATA RECORD DATA-IN.
01  DATA-IN          PIC X(80).
LINKAGE SECTION.
01  CONTROL-FIELD    PIC XXX.
01  RECORD-IN        PIC X(21).
PROCEDURE DIVISION USING CONTROL-FIELD RECORD-IN.
BEGIN.
  GO TO FILE-OPEN.
FILE-OPEN.
  OPEN INPUT USRFILE
  MOVE SPACES TO CONTROL-FIELD.
  ALTER BEGIN TO PROCEED TO FILE-READ.
FILE-READ.
  READ USRFILE INTO RECORD-IN
  AT END
  MOVE 'END' TO CONTROL-FIELD
  CLOSE USRFILE
  ALTER BEGIN TO PROCEED TO FILE-OPEN.
GOBACK.
```

# 17 CALL LOOP

---

▪ 機能 .....	106
▪ 制限事項 .....	106
▪ 構文説明 .....	107
▪ 例 .....	107

ストラクチャードモード構文

```
CALL LOOP operand1 [operand2] ...40  
statement ...  
END-LOOP
```

レポーティングモード構文

```
CALL LOOP operand1 [operand2] ...40  
statement ...  
[LOOP]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CALL](#) | [CALL FILE](#) | [CALLNAT](#) | [DEFINE SUBROUTINE](#) | [ESCAPE](#) | [FETCH](#) | [PERFORM](#)

関連機能グループ：「[プログラムおよびルーチンの呼び出し](#)」

## 機能

---

CALL LOOP ステートメントは、Natural 以外のプログラムを呼び出す処理ループを生成します。

CALL ステートメントとは異なり、CALL LOOP ステートメントは、処理ループ内で繰り返し Natural 以外のプログラムを呼び出します。CALL 処理の詳細については、[CALL ステートメント](#)を参照してください。

## 制限事項

---

ステートメント [AT BREAK](#)、[AT START OF DATA](#)、および [AT END OF DATA](#) を、CALL LOOP 処理ループ内で使用することはできません。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	動的定義
<i>operand1</i>	C S	A	可	不可
<i>operand2</i>	C S A G	A U N P I F B D T L C	可	可

構文要素の説明：

<i>operand1</i>	呼び出される Natural プログラムの名前は、定数または変数で指定できます。プログラムロジックによって呼び出すプログラムが異なる場合は、長さ1~8の英数字の変数として指定します。変数では、プログラム名を左詰めで指定する必要があります。
<i>operand2</i>	CALL LOOP ステートメントは、最大 40 個までのパラメータを受け渡すことができます。パラメータリストの構成は、CALL ステートメントと同じです。パラメータリストのフィールドは、CALL LOOP ステートメント自体で最初に定義するか、または事前に定義しておくことができます。
<i>statement ...</i>	CALL LOOP ステートメントによって開始された処理ループは、ESCAPE ステートメントで終了する必要があります。
<b>END-LOOP</b>	Natural 予約語 <b>END-LOOP</b> を使用して、処理ループを閉じる必要があります。

## 例

```

DEFINE DATA LOCAL
1 PARAMETER1 (A10)
END-DEFINE
CALL LOOP 'ABC' PARAMETER1
  IF PARAMETER1 = 'END'
    ESCAPE BOTTOM
  END-IF
END-LOOP
END


```

---

# 18 CALLNAT

---

▪ 機能 .....	110
▪ 構文説明 .....	111
▪ ダイナミック変数を使用したパラメータ引き渡し .....	113
▪ 例 .....	114



構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CALL](#) | [CALL FILE](#) | [CALL LOOP](#) | [DEFINE SUBROUTINE](#) | [ESCAPE](#)  
| [FETCH](#) | [PERFORM](#)

関連機能グループ：「[プログラムおよびルーチンの呼び出し](#)」

## 機能

CALLNAT ステートメントは、Natural サブプログラムを実行するために呼び出します。Natural サブプログラムは、CALLNAT ステートメントによってのみ呼び出すことができます。サブプログラムを単独で実行することはできません。

CALLNAT ステートメントを実行すると、呼び出し元オブジェクト（CALLNAT ステートメントを含んでいるオブジェクト）は中断され、呼び出したサブプログラムが実行されます。サブプログラムの実行は、END ステートメントに到達するまで、または [ESCAPE ROUTINE](#) ステートメントの実行によってサブプログラムの処理が停止されるまで続きます。どちらの場合でも、呼び出し元オブジェクトのその後の処理は、CALLNAT ステートメントの次にあるステートメントから継続されます。

### 注意:

1. サブプログラムは、他のサブプログラムを交互に呼び出すことができます。
2. サブプログラムには、呼び出し側オブジェクトが使用するグローバルデータエリアへのアクセスがありません。サブプログラムからサブルーチンやヘルプルーチンを呼び出す場合は、サブプログラムで独自のグローバルデータエリアを作成して、サブルーチンまたはヘルプルーチンと共有することができます。



## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	動的定義
<i>operand1</i>	C S	A	可	不可
<i>operand2</i>	C S A G	A U N P I F B D T L C G O	可	可

構文要素の説明：

<i>operand1</i>	<p>サブプログラム名：</p> <p><i>operand1</i> として、呼び出すサブプログラムの名前を指定します。名前は、1~8 文字の定数または変数で指定できます。プログラムロジックによって呼び出すサブプログラムが異なる場合は、長さ 1~8 の英数字の変数として指定します。</p> <p>サブプログラム名に、アンパサンド (&amp;) を含めることができます。これは、実行時にシステム変数 *LANGUAGE の現在の値で置き換えられます。これにより、入力時の言語に応じて、入力処理ごとに異なるサブプログラムを呼び出すことができます。</p>
<i>operand2</i>	<p>パラメータ：</p> <p>パラメータをサブプログラムに渡す場合、パラメータリストの構成は、<b>DEFINE DATA PARAMETER</b> ステートメントで定義する必要があります。CALLNAT ステートメントで指定したパラメータだけが、呼び出し元オブジェクトからサブプログラムに有効なデータです。</p> <p>デフォルトでは、パラメータは参照によって渡されます。つまり、データはアドレスパラメータを介して転送され、パラメータ値自体は移動されません。ただし、パラメータを値によって渡す、つまり実際のパラメータ値を渡すこともできます。これを行うには、サブプログラムの <b>DEFINE DATA PARAMETER</b> ステートメントで、これらのフィールドを <b>BY VALUE</b> または <b>BY VALUE RESULT</b> オプション付きで定義します (<b>DEFINE DATA</b> ステートメントの説明にある <i>parameter-data-definition</i> を参照)。</p> <ul style="list-style-type: none"> <li>■ パラメータを参照によって渡す場合は、次のことが適用されます。呼び出し元オブジェクトのパラメータの順番、フォーマット、および長さは、呼び出されるサブプログラムの <b>DEFINE DATA PARAMETER</b> 構造の順番、フォーマット、および長さとは正確に一致させる必要があります。呼び出し元オブジェクトの変数名と呼び出されるサブプログラムの変数名は、異なってもかまいません。</li> <li>■ パラメータを値によって渡す場合は、次のことが適用されます。呼び出し元オブジェクトのパラメータの順番は、呼び出されるサブプログラムの <b>DEFINE DATA PARAMETER</b> 構造の順番と正確に一致させる必要があります。呼び出し元オブジェクトの変数のフォーマットと長さは、サブプログラムの変数と異なってもかまいません。ただし、データ転送の互換性が必要です (詳細については、『プログラミングガイド』の「演算割り当てのルール」、「データ転送」にある対応表を参照)。呼び出し元オブジェクトと呼び出されるサブプログラムで異なる変数名を使用できます。サブプログラムで変更されたパラメータの値を呼び出し元オブジェクトに戻す場合、これらのフィールドを <b>BY VALUE RESULT</b> 付き</li> </ul>

	<p>で定義する必要があります。BY VALUE では (RESULT を使用せずに)、修正されたパラメータ値を呼び出し側オブジェクトに戻すことはできません (AD 指定とは関係ありません。以下も参照してください)。</p> <p><b>注意:</b> BY VALUE では、パラメータ値の内部コピーが作成されます。サブプログラムがこのコピーにアクセスして、その値を変更しても、呼び出し元オブジェクトのオリジナルのパラメータ値には影響しません。BY VALUE RESULT では、同様に内部的なコピーが作成されます。ただし、サブプログラム終了後、オリジナルのパラメータ値はコピーの値 (変更された) で上書きされます。</p> <p>いずれのパラメータ引き渡し方法にも、次が適用されます。</p> <p><i>operand2</i> としてグループを指定すると、グループ内にある個々のフィールドがサブプログラムに渡されます。つまり、これらの各フィールドに対応するフィールドを、サブプログラムのパラメータデータエリアに定義する必要があります。</p> <p>呼び出されたサブプログラムのパラメータデータエリアでは、グループの再定義は REDEFINE ブロック内だけで行うことができます。</p> <p>配列を渡す場合、サブプログラムのパラメータデータエリアの次元数とオカレンス数は CALLNAT パラメータリストと同じにする必要があります。</p> <p><b>注意:</b> 添字付きグループの一部として定義された配列の複数オカレンスを CALLNAT ステートメントで渡す場合は、サブプログラムのパラメータデータエリア内の対応するフィールドを再定義することはできません。再定義すると、不正なアドレスが渡されます。</p> <p>COMPOPT コマンドのオプション PCHECK を ON に設定すると、CALLNAT ステートメントで指定されたパラメータの数値、フォーマット、長さ、および配列の添字の限度がコンパイラによってチェックされます。また、DEFINE DATA PARAMETER ステートメントの OPTIONAL 機能もパラメータチェックで考慮されます。</p>						
AD=	<p>属性定義：</p> <p><i>operand2</i> が変数の場合は、次のいずれかの方法でマークすることができます。</p> <table border="1" data-bbox="272 1297 1385 1774"> <tr> <td data-bbox="272 1297 808 1554">AD=O</td> <td data-bbox="808 1297 1385 1554"> <p>変更不可。セッションパラメータ AD=O を参照してください。</p> <p><b>注意:</b> 内部的には、AD=O は BY VALUE と同様に処理されます (DEFINE DATA ステートメントの説明にある <i>parameter-data-definition</i> を参照)。</p> </td> </tr> <tr> <td data-bbox="272 1554 808 1696">AD=M</td> <td data-bbox="808 1554 1385 1696"> <p>変更可。セッションパラメータ AD=M を参照してください。</p> <p>これはデフォルト設定です。</p> </td> </tr> <tr> <td data-bbox="272 1696 808 1774">AD=A</td> <td data-bbox="808 1696 1385 1774"> <p>入力のみ。セッションパラメータ AD=A を参照してください。</p> </td> </tr> </table> <p><i>operand2</i> が定数の場合は、AD を明示的に指定することはできません。定数には常に AD=O が適用されます。</p>	AD=O	<p>変更不可。セッションパラメータ AD=O を参照してください。</p> <p><b>注意:</b> 内部的には、AD=O は BY VALUE と同様に処理されます (DEFINE DATA ステートメントの説明にある <i>parameter-data-definition</i> を参照)。</p>	AD=M	<p>変更可。セッションパラメータ AD=M を参照してください。</p> <p>これはデフォルト設定です。</p>	AD=A	<p>入力のみ。セッションパラメータ AD=A を参照してください。</p>
AD=O	<p>変更不可。セッションパラメータ AD=O を参照してください。</p> <p><b>注意:</b> 内部的には、AD=O は BY VALUE と同様に処理されます (DEFINE DATA ステートメントの説明にある <i>parameter-data-definition</i> を参照)。</p>						
AD=M	<p>変更可。セッションパラメータ AD=M を参照してください。</p> <p>これはデフォルト設定です。</p>						
AD=A	<p>入力のみ。セッションパラメータ AD=A を参照してください。</p>						

$nX$	<p>表記 <math>nX</math> を使用して、次の <math>n</math> 個のパラメータをスキップするように指定することができます（例えば、<math>1X</math> を使用すると次のパラメータがスキップされ、<math>3X</math> を使用すると次の3つのパラメータがスキップされます）。これは、次の <math>n</math> 個のパラメータでは、値がサブプログラムに渡されないことを意味します。 <math>n</math> の値の有効範囲は 1~4096 です。</p> <p>スキップするパラメータは、サブプログラムの <code>DEFINE DATA PARAMETER</code> ステートメント内のキーワード <code>OPTIONAL</code> を使用して定義する必要があります。 <code>OPTIONAL</code> は、値を呼び出し側オブジェクトからこのようなパラメータに渡すこともできるということを意味します。</p>
------	--

## ダイナミック変数を使用したパラメータ引き渡し

ダイナミック変数は、呼び出されたプログラムオブジェクト（CALLNAT、PERFORM）へのパラメータとして渡すことができます。ダイナミック変数の値スペースは連続しているので、参照渡しが可能です。値渡しを使用すると、呼び出し元の変数定義がソースオペランドとして割り当てられ、パラメータ定義が応答先オペランドとして割り当てられます。さらに、値による呼び出しは、結果として反対方向への動作を変更します。参照による呼び出しを使用する場合、変数定義およびパラメータ定義は `DYNAMIC` である必要があります。そのうちの1つだけが `DYNAMIC` の場合、ランタイムエラーが発生します。値による呼び出し（の結果）の場合は、すべての組み合わせが可能です。

次の表は、パラメータの転送に関して、呼び出し元のスタティックおよびダイナミックに定義された変数と、スタティックおよびダイナミックに定義されたパラメータの有効な組み合わせを示しています。


### 参照渡し

呼び出し元の operand2	パラメータ定義	
	スタティック	ダイナミック
スタティック	○	×
ダイナミック	×	○

ダイナミック変数 A または B のフォーマットは一致している必要があります。

### 値渡し（結果）

呼び出し元の operand2	パラメータ定義	
	スタティック	ダイナミック
スタティック	○	○
ダイナミック	○	○

 **注意:** スタティック/ダイナミックまたはダイナミック/スタティック定義を使用する場合は、割り当ての際にデータ転送の規則によって値が切り捨てられることがあります。

## 例

---

- 例 1
- 例 2

### 例 1

呼び出す側のプログラム：

```
** Example 'CNTEX1': CALLNAT
*****
DEFINE DATA LOCAL
1 #FIELD1 (N6)
1 #FIELD2 (A20)
1 #FIELD3 (A10)
END-DEFINE
*
CALLNAT 'CNTEX1N' #FIELD1 (AD=M) #FIELD2 (AD=0) #FIELD3 'P4 TEXT'
*
WRITE '=' #FIELD1 '=' #FIELD2 '=' #FIELD3
*
END
```

呼び出される側のサブプログラム **CNTEX1N**：

```
** Example 'CNTEX1N': CALLNAT (called by CNTEX1)
*****
DEFINE DATA PARAMETER
1 #FIELDA (N6)
1 #FIELDB (A20)
1 #FIELD C (A10)
1 #FIELD D (A7)
END-DEFINE
*
*
#FIELD A := 4711
*
#FIELD B := 'HALLO'
*
#FIELD C := 'ABC'
*
WRITE '=' #FIELD A '=' #FIELD B '=' #FIELD C '=' #FIELD D
*
END
```

## 例 2

呼び出す側のプログラム：

```
** Example 'CNTEX2': CALLNAT
*****
DEFINE DATA LOCAL
1 #ARRAY1 (N4/1:10,1:10)
1 #NUM (N2)
END-DEFINE
*
*
CALLNAT 'CNTEX2N' #ARRAY1 (2:5,*)
*
FOR #NUM 1 TO 10
  WRITE #NUM #ARRAY1(#NUM,1:10)
END-FOR
*
END
```

呼び出される側のサブプログラム CNTEX2N：

```
** Example 'CNTEX2N': CALLNAT (called by CNTEX2)
*****
DEFINE DATA
PARAMETER
1 #ARRAY (N4/1:4,1:10)
LOCAL
1 I (I2)
END-DEFINE
*
*
FOR I 1 10
  #ARRAY(1,I) := I
  #ARRAY(2,I) := 100 + I
  #ARRAY(3,I) := 200 + I
  #ARRAY(4,I) := 300 + I
END-FOR
*
END
```



# 19 CLOSE CONVERSATION

---

■ 機能 .....	118
■ 構文説明 .....	118
■ 詳細と例 .....	119

CLOSE CONVERSATION	{	{operand1} ...	}
		*CONVID	
		ALL	

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[DEFINE DATA CONTEXT](#) | [OPEN CONVERSATION](#)

関連機能グループ：「[Natural リモートプロシージャコール](#)」

## 機能

CLOSE CONVERSATION ステートメントは、Natural RPC と一緒に使用します。これにより、クライアントは会話を閉じることができます。現在の会話、開かれた別の会話、または開かれたすべての会話を閉じることができます。



**注意:** 別のライブラリにログオンしても、会話が自動的に閉じることはありません。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A	I	可	不可

構文要素の説明：

<i>operand1</i>	閉じられる会話： 開かれている特定の会話を閉じるには、その ID を <i>operand1</i> として指定します。 <i>operand1</i> はフォーマット／長さが I4 の変数にする必要があります。
*CONVID	現在の会話を閉じるには、*CONVID を指定します。現在の会話の ID は、システム変数 *CONVID の値によって決定されます。
ALL	開かれているすべての会話を閉じる場合は、ALL を指定します。



## 詳細と例

---

『*Natural* リモートプロシージャコール (RPC) 』ドキュメントの次のセクションを参照してください。

- 会話型モードでの *Natural* RPC の動作
- 会話型 RPC の使用



# 20 CLOSE PC FILE

---

■ 機能 .....	122
■ 構文説明 .....	122
■ 例 .....	122

<code>CLOSE { PC WORK } [FILE] work-file-number</code>
--

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[DOWNLOAD PC FILE](#) | [UPLOAD PC FILE](#)

関連機能グループ：「[ワークファイル/PC ファイルの制御](#)」

## 機能

---

CLOSE PC FILE ステートメントは、特定の PC ワークファイルをクローズします。これにより、クローズする PC ワークファイルをプログラム内で指定できます。

また、コマンドモードになった場合は、ワークファイルが自動的にクローズされます。

NETWORK マクロの設定が適用されます。

詳細については、『[Natural Connection](#)』ドキュメントと『[Entire Connection](#)』ドキュメントを参照してください。

## 構文説明

---

<code>work-file-number</code>	<code>work-file-number</code> は、クローズする PC ワークファイルの番号です。 この番号は、Natural に対する定義に従って、PC のワークファイル番号の 1 つに対応している必要があります。
-------------------------------	---

## 例

---

次のプログラムは、CLOSE PC FILE ステートメントの使用方法を示しています。

```
** Example 'PCCLEX1': CLOSE PC FILE
**
** NOTE: Example requires that Natural Connection is installed.
*****
DEFINE DATA LOCAL
01 W-DAT (A40)
01 REC-NUM (N3)
01 I (P3)
END-DEFINE
```

```
*
REPEAT
UPLOAD PC FILE 7 ONCE W-DAT          /* Data upload
  AT END OF FILE
  ESCAPE BOTTOM
  END-ENDFILE
  INPUT 'Processing file' W-DAT (AD=0)
  /  'Enter record number to display' REC-NUM
  IF REC-NUM = 0
  STOP
  END-IF
  FOR I = 1 TO REC-NUM
  UPLOAD PC FILE 7 ONCE W-DAT
  AT END OF FILE
  WRITE 'Max. record number reached, last record is'
  ESCAPE BOTTOM
  END-ENDFILE
  END-FOR
  I := I - 1
  WRITE 'Record' I ':' W-DAT
CLOSE PC FILE 7                      /* Close PC file 7
END-REPEAT
END
```

プログラム **PCCLEX1** の出力：

プログラムを実行するとウィンドウが開くので、そこでアップロードするデータが含まれている PC ファイルの名前を指定します。その後、データが PC からアップロードされます。各ループの最後で、PC ワークファイルがクローズされます。



# 21 CLOSE PRINTER

---

■ 機能 .....	126
■ 構文説明 .....	126
■ 例 .....	127

```
CLOSE PRINTER { (logical-printer-name)
                (printer-number) }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT END OF PAGE](#) | [AT TOP OF PAGE](#) | [DEFINE PRINTER](#) | [DISPLAY](#) | [EJECT](#) | [FORMAT](#) | [NEWPAGE](#) | [PRINT](#) | [SKIP](#) | [SUSPEND IDENTICAL SUPPRESS](#) | [WRITE](#) | [WRITE TITLE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

CLOSE PRINTER ステートメントは、特定のプリンタを閉じます。このステートメントを使用して、閉じるプリンタをプログラム内で明示的に指定します。

また、次のいずれかの場合は、プリンタが自動的に閉じられます。

- 同じプリンタをもう一度定義する [DEFINE PRINTER](#) ステートメントが実行された場合。
- コマンドモードになった場合。

## 構文説明

<i>logical-printer-name</i>	<p>論理プリンタ名：</p> <p><i>logical-printer-name</i> で、どのプリンタを閉じるかを指定します。このプリンタ名およびプリンタ番号は、対応する <a href="#">DEFINE PRINTER</a> ステートメントで定義したプリンタと同じものです。</p> <p><i>logical-printer-name</i> の命名規則はユーザー定義変数と同じ規則です。詳細については、『<a href="#">Natural の使用</a>』ドキュメントの「<a href="#">ユーザー定義変数の命名規則</a>」を参照してください。</p>
<i>printer-number</i>	<p>プリンタ番号：</p> <p><i>logical-printer-name</i> の代わりに、<i>printer-number</i> を定義して閉じるプリンタを指定することもできます。</p> <p><i>printer-number</i> は、0～31 の範囲にある番号になります。この番号は、<a href="#">DISPLAY/WRITE</a> または <a href="#">DEFINE PRINTER</a> ステートメントでも使用します。</p> <p>プリンタ番号 0 はハードコピープリンタを示します。</p>



## 例

```
** Example 'CLPEX1': CLOSE PRINTER
*****
DEFINE DATA LOCAL
1 EMP-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
  2 BIRTH
*
1 #I-NAME (A20)
END-DEFINE
*
DEFINE PRINTER (PRT01=1)
*
REPEAT
  INPUT 'SELECT PERSON' #I-NAME
  IF #I-NAME = ' '
    STOP
  END-IF
  FIND EMP-VIEW WITH NAME = #I-NAME
  WRITE (PRT01) 'NAME           :' NAME ',' FIRST-NAME
              /      'PERSONNEL-ID :' PERSONNEL-ID
              /      'BIRTH           :' BIRTH (EM=YYYY-MM-DD)
  END-FIND
/*
  CLOSE PRINTER (PRT01)
/*
END-REPEAT
END
```



# 22 CLOSE WORK FILE

---

■ 機能 .....	130
■ 構文説明 .....	130
■ 例 .....	130

CLOSE WORK [FILE] *work-file-number*

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[DEFINE WORK FILE](#) | [READ WORK FILE](#) | [WRITE WORK FILE](#)

関連機能グループ：「[ワークファイル／PC ファイルの制御](#)」

## 機能

---

CLOSE WORK FILE ステートメントは、特定のワークファイルをクローズします。これにより、クローズするワークファイルをプログラム内で指定できます。

次の場合は、ワークファイルが自動的にクローズされます。

- コマンドモードになった場合。
- [READ WORK FILE](#) ステートメントの実行中にエンドオブファイル条件が発生した場合。
- 別のデータセットを関連するワークファイル番号に割り当てる [DEFINE WORK FILE](#) ステートメントが実行される前。
- プロファイルパラメータ WORK のサブパラメータ CLOSE により。

プロファイルパラメータ WORK で CLOSE=FIN が指定されたワークファイルに対する CLOSE WORK FILE ステートメントは無視されます。

## 構文説明

---

<i>work-file-number</i>	閉じるワークファイルの番号（Natural に定義した）。
-------------------------	-------------------------------

## 例

---

```
** Example 'CWFEX1': CLOSE WORK FILE
*****
DEFINE DATA LOCAL
1 W-DAT (A20)
1 REC-NUM (N3)
1 I (P3)
END-DEFINE
*
REPEAT
```

```
READ WORK FILE 1 ONCE W-DAT /* READ MASTER RECORD
/*
AT END OF FILE
  ESCAPE BOTTOM
END-ENDFILE
INPUT 'PROCESSING FILE' W-DAT (AD=0)
  / 'ENTER RECORDNUMBER TO DISPLAY' REC-NUM
IF REC-NUM = 0
  STOP
END-IF
  FOR I = 1 TO REC-NUM
  /*
  READ WORK FILE 1 ONCE W-DAT
  /*
  AT END OF FILE
    WRITE 'RECORD-NUMBER TOO HIGH, LAST RECORD IS'
    ESCAPE BOTTOM
  END-ENDFILE
END-FOR
I := I - 1
WRITE 'RECORD' I ':' W-DAT
/*
CLOSE WORK FILE 1
/*
END-REPEAT
END
```



# 23 COMPOSE

---

■ 機能 .....	134
■ 構文説明 .....	135
■ フォーマット処理 .....	145
■ ダイアログモード .....	146
■ Natural 以外のプログラム .....	147
■ 例 .....	149

このステートメントは、Con-nect または Con-form がインストールされている場合にのみ使用できます。

```
COMPOSE
[RESETTING-clause]
[MOVING-clause]
[ASSIGNING-clause]
[FORMATTING-clause]
[EXTRACTING-clause]
```

複数の節を指定した場合、これらの節は上記の順序で処理されます。

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

---

COMPOSE ステートメントでは、Con-form (Con-nect 内のテキストフォーマッタ) によるテキストのフォーマットを、Natural プログラムから直接開始します。

フォーマットするテキストは、変数を使用して提供するか、または Con-nect テキストブロック (Con-form フォーマットコマンドを含むドキュメント) から取得できます。

Natural 変数の内容は、フォーマットするテキストにダイナミックに組み込む変数として Con-form に渡すことができます。

また、Con-form 変数に含まれている値は、テキストフォーマッタから Natural プログラムに返すこともできます。

Con-form の手続きが完了すると (ドキュメントがフォーマットされると)、出力は次の場所のいずれかに渡されます。

- Natural レポート
- Con-nect システムファイル内のドキュメント
- COMPOSE ステートメントを実行する Natural プログラム内の変数
- Natural 以外のプログラム



## 構文説明


<i>RESETTING-clause</i>	この節は、テキストフォーマットバッファエリアからの情報を削除して、CSIZEによって割り当てられた COMPOSE バッファからメモリを解放します。「 <b>RESETTING 節</b> 」を参照してください。
<i>MOVING-clause</i>	この節は、テキスト行をテキストフォーマッタバッファエリアに移動、またはフォーマッタに直接移動して、フォーマットされたテキスト出力をフォーマッタのワークスペースから取得します。「 <b>MOVING 節</b> 」を参照してください。
<i>ASSIGNING-clause</i>	この節は、Natural 変数をテキスト変数に割り当てます。「 <b>ASSIGNING 節</b> 」を参照してください。
<i>FORMATTING-clause</i>	この節は、テキストおよび Con-form ステートメントを組み合わせた入力を使用して、最終的なフォーマット形式（正しい改行と改ページを含む）でテキストを作成します。「 <b>FORMATTING 節</b> 」を参照してください。
<i>EXTRACTING-clause</i>	この節は、テキスト変数の値を Natural 変数に割り当てます。「 <b>EXTRACTING 節</b> 」を参照してください。

### RESETTING 節

RESETTING	<table border="1"> <tr> <td>DATAAREA</td> </tr> <tr> <td>TEXTAREA</td> </tr> <tr> <td>MACROAREA</td> </tr> <tr> <td>ALL</td> </tr> </table>	DATAAREA	TEXTAREA	MACROAREA	ALL
DATAAREA					
TEXTAREA					
MACROAREA					
ALL					

この節は、テキストフォーマットバッファエリアから次のものを削除します。

- DATAAREA は、すべてのアクティブなテキスト変数を削除します。
- TEXTAREA は、すべてのテキスト入力データを削除します。
- MACROAREA は、すべてのテキストマクロを削除します。
- ALL は、上記のすべてを削除します。

 **注意:** 互換性に関する理由から、キーワード TEXTAREA は、MOVING 節で使用されるものと同一フォーマッタの「データエリア」を表します。

## MOVING 節

この節は、1つ以上のテキスト値をテキストフォーマットバッファエリアに移動します（「[構文 1](#)」）。このエリアは、フォーマット操作の入力ソースとして使用できます。テキストフォーマッタが現時点で入力待ちになっている場合（「[ダイアログモード](#)」を参照）、テキストは Con-form のテキストエリアには格納されずに、フォーマッタに直接渡されます（「[Syntax 1](#)」と「[構文 2](#)」）。ソース入力は、LAST オプションで終了します。フォーマットされたテキストが現時点で出力待ちになっている場合は（「[ダイアログモード](#)」を参照）、MOVING 節の [構文 3](#) を使用して、Natural プログラムからフォーマッタに制御を戻します。ステータス変数の詳細については、「[FORMATTING 節](#)」を参照してください。

ダイアログモードのステータスに応じて、MOVING 節を次の形式のいずれかで使用できます。

### 構文 1

MOVING 節の [構文 1](#) は、フォーマットがまだ始まっていない場合、またはフォーマッタが入力用のダイアログモードで入力（最初のステータス変数である TERM）を待っている場合に適しています。

```
MOVING [operand1] ... 37 [TO DATAAREA] [LAST] [STATUS [TO] operand2 [operand3 [operand4 [operand5]]]]
```

### 構文 2

MOVING 節の [構文 2](#) は、フォーマッタが入出力両用のダイアログモードで追加の入力（最初のステータス変数である TERM）を待っている場合に適しています。このモードでは、フォーマッタは複数行の入力を受け入れません。

実行コンテキストは、一連の COMPOSE ステートメントを実行する間に変わることがあります。したがって、フォーマッタが入力待ちの場合でも、出力変数を再指定する必要があります。

```
MOVING { [operand1] [TO DATAAREA] } [OUTPUT] TO
        LAST                               VARIABLES operand6 ... 20
        [STATUS [TO] operand2 [operand3 [operand4 [operand5]]]]
```

### 構文 3

MOVING 節の構文 3 は、フォーマッタが出力用（および同時に入力用の可能性もあり）のダイアログモードで、出力（最初のステータス変数内である STRG）を Natural プログラムに渡す場合に適しています。

```
MOVINGOUTPUT [TO VARIABLES] operand6 ... 20 [STATUS [T0] operand2 [operand3 [operand4 [operand5]]]]
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S A	A N P	可	不可
<i>operand2</i>	S	A	可	可
<i>operand3</i>	S	B	可	可
<i>operand4</i>	S	B	可	可
<i>operand5</i>	S	B	可	可
<i>operand6</i>	S A	A	可	不可

*operand2* は、フォーマット／長さ A4 で定義する必要があります。*operand3*、*operand4*、および *operand5* は、フォーマット／長さ B4 で定義する必要があります。

### ASSIGNING 節

この節は、値を Con-form テキスト変数に割り当てます。これらのテキスト変数は、フォーマット操作の中で後で参照できます。

テキスト変数名は、大文字で指定する必要があります。

```
ASSIGNING [TEXTVARIABLE] {operand1=operand2}, ... 19
```


オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S	A	可	不可
<i>operand2</i>	C S	A N P	可	可

## FORMATTING 節

この節では、フォーマットされた出力が Con-form によって生成されます。

フォーマットオプションは、1つ以上の副節で指定します。副節を省略すると、Con-form によってデフォルトのフォーマットオプションが適用されます。ステータス変数は、ダイアログモードで使用されます。



構文要素の説明：

<i>OUTPUT-subclause</i>	出力メディア。Natural レポート、Con-nect キャビネット、1つ以上の Natural 変数（または Natural 変数の配列）、または Natural 以外のプログラムを、出力メディアにすることができます。「 <b>OUTPUT 副節</b> 」を参照してください。	
<i>INPUT-subclause</i>	入力メディア。Con-nect ドキュメント、COMPOSE データエリア（「 <b>MOVING 節</b> 」を参照）、COMPOSE ステートメントを実行する Natural プログラムの環境（「 <b>MOVING 節</b> 」を参照）、Natural プログラム、またはこれらの4つを混合したものを、入力メディアにすることができます。	
<i>STATUS-subclause</i>	フォーマット操作のステータス。フォーマット操作では、COMPOSE ステートメントを複数回実行することがあります（ <b>ダイアログモード</b> ）。例えば、入力は Natural プログラムによってフォーマッタのワークスペースに送られ、出力はフォーマッタのワークスペースから Natural プログラムの環境（1つ以上の Natural 変数）に渡されます。したがって、フォーマット操作のステータスを Natural プログラムに通知する必要があります。	
	フォーマット処理の間に、次の変数が Natural プログラムに渡されます。	
	<b>State</b>	ダイアログモードで入力の準備ができている場合は TERM。
		ダイアログモードで出力の準備ができている場合は STRG。
		フォーマット処理が正常に完了した場合は END。
		フォーマット処理が正常に完了しなかった場合は ENDX。

	<b>Position</b>	フォーマットしているドキュメントのページ番号と行番号。ページ番号と行番号は、2つの変数（ページ位置と行位置）で別々に保持されます。
	<b>Amount of Output Data</b>	Natural プログラムに渡されるフォーマット済み出力の行数。フォーマッタは、値が挿入される次の出力変数へのポインタとしてこの番号を使用します。この値は、出力行が発行されると1ずつ増加します 現在の値が範囲外にある場合、値は1に設定されます。
<i>PROFILE-subclause</i>	入力処理の前に処理されるテキストブロック。	
<i>MESSAGES-subclause</i>	警告メッセージと統計情報の出力、およびエラー処理を制御します。	
<i>ERRORS-subclauses</i>		
<i>ENDING-subclause</i>	「 <a href="#">ENDING 副節</a> 」を参照してください。	
<i>STARTING-subclause</i>	「 <a href="#">STARTING 副節</a> 」を参照してください。	

## OUTPUT 副節

この副節を使用すると、Con-form でフォーマットされたテキストの出力先を指定できます

この副節を省略すると、デフォルトの出力デバイスとしてNaturalのメインプリンタが使用されます。

OUTPUT	{	( <i>rep</i> )	}
		SUPPRESSED	
		CALLING <i>operand1</i>	
		TO VARIABLES [CONTROL <i>operand2 operand3</i> ] <i>operand4</i> ... 20	
		DOCUMENT- <i>option</i>	

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S	A	可	不可
<i>operand2</i>	C S	A	可	不可
<i>operand3</i>	C S	A	可	不可
<i>operand4</i>	S A	A	可	不可

## 構文要素の説明：

<b>OUTPUT</b> ( <i>rep</i> )	<p>出力先をプリンタに指定し（レポート番号が0ではない）、Connect プリンタプロファイルが Connect API 関数 Z-DRIVER によってすでにロードされている場合、そのプロファイルの設定は、フォーマットされる出力テキストのテキスト強調表示オプションを制御するために使用されます。</p> <p>プリンタプロファイルがアクティブであり、論理フォームフィードの制御を指定していなかった場合は、Natural ニュークリアスの適切な内部機能を使用して改ページが挿入されます。</p> <p>現在アクティブになっている Connect プロファイルに反映されていないその他のテキスト強調表示オプションは、すべて無視されます。</p> <p><b>注意:</b> レポート以外の出力先を指定した COMPOSE RESETTNG ALL または COMPOSE FORMATTING ステートメントを実行すると、フォーマッタのワークスペースからプリンタプロファイルがアンロードされます。</p> <p>出力先をレポート 0 に指定するか、またはプリンタプロファイルがアクティブでない場合は、出力処理の役割が Connect から Natural ニュークリアスルーチンに渡されます。この状況では、太字、下線、および斜体といったテキスト強調表示オプションのみが認識されます。</p> <p><b>注意:</b> DEFINE PRINTER (<i>n</i>) OUTPUT 'CONNECT' ステートメントで参照されるレポートは、COMPOSE FORMATTING ステートメントで出力先に指定することはできません。</p>
<b>OUTPUT SUPPRESSED</b>	このオプションを指定すると、出力が SUPPRESSED されます。
<b>OUTPUT CALLING</b>	「 <a href="#">Natural 以外のプログラム</a> 」セクションを参照してください。
<b>OUTPUT TO VARIABLES</b>	<p>通常、フォーマットしたテキストは、最終的なフォーマット形式で Natural 変数の配列に渡されます。各行は変数に格納されます（必要に応じて、変数に適合するように行は切り捨てられます）。テキスト強調表示オプションは無視されます。ただし、例外として、CONTROL 変数で指定した強調表示オプションは、テキストの一部分の強調に使用されます（太字または下線）。</p> <p>CONTROL 変数、I、および N を指定すると、中間フォーマット形式（分散論理制御シーケンスを含む）でフォーマットしたテキストが生成されます。</p> <p><i>operand2</i> と <i>operand3</i> は、フォーマット/長さを A1 にする必要があります。</p> <p>詳細については、「<a href="#">ダイアログモード</a>」を参照してください。</p>
<b>OUTPUT DOCUMENT</b>	下記の「 <a href="#">DOCUMENT オプション</a> 」を参照してください。

## DOCUMENT オプション

DOCUMENT	[	INTQ	[	{	FINAL	}	]	[CABINET]	]
					INTERMEDIATE			<i>operand1</i>	
								[PASSW= <i>operand2</i> ]	
	[	[GIVING]	{	<i>operand3</i> [ <i>operand4</i> ]	}				
				<i>operand4</i> [ <i>operand3</i> ]					

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S	A	可	不可
<i>operand2</i>	S	A	可	不可
<i>operand3</i>	S	B	可	可
<i>operand4</i>	S	B	可	可

## OUTPUT DOCUMENT

*operand3*（フォーマット／長さB10）はフォーマッタによって使用され、ユニークキーをドキュメントからNaturalプログラムに戻します。このオプションは、互換性保持のためにのみサポートされています。

*operand4*（フォーマット／長さB4）はフォーマッタによって使用され、フォーマットされた出力ドキュメントを指すISNをNaturalプログラムに戻します。このISNは、Con-nect APIの連続的な呼び出しでドキュメントを参照する場合に有用です。

*operand1*（長さは最大8文字まで）を指定しなかった場合、ドキュメントは現在のユーザーのキャビネットに追加されます（現在アクティブなNaturalユーザーIDと同じIDを持つキャビネットに追加）。

現在想定されているユーザーIDでアクセスできないキャビネットにドキュメントを保存する場合は、パスワード（最大8文字まで）を指定する必要があります。


Con-formはCon-nectアクセス制限の準拠を強化し、Con-nectに定義済みのキャビネットIDだけを受け入れます。

 **注意:** キャビネットIDは、大文字で指定する必要があります。

ドキュメントは、ドキュメント名なしでCOMPOSEフォルダに追加されます。タイトル行には、COMPOSE FORMATTINGステートメントを実行するプログラムの名前が、実行の日時とともに挿入されます。

キーワード `INTERMEDIATE` を省略すると、ドキュメントは最終的なフォーマット形式のテキストで作成されます。この場合、太字や斜体などの特定のテキスト強調表示オプションは無視されます。

**INPUT 副節**



オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S	A	可	不可
<i>operand2</i>	C S	A	可	不可
<i>operand3</i>	S	A	可	不可

この副節は、テキストフォーマッタに入力を提供するソースを指定します。入力は、1つ以上の `MOVING` 操作によって値を挿入する必要がある `Con-form` のデータエリア（データエリアとダイアログモードからのテキストを組み合わせることも可能）から取得されるか、またはテキストブロック（*operand1*で指定）から取得されます。テキストブロックは、`Con-nect` キャビネットに含まれているか、または **Natural 以外のプログラム** によって提供されます。テキストブロックは、`CALL` ステートメントに適用されるものと同じ構文を使用して呼び出します。`Con-nect` キャビネットまたは **Natural 以外のプログラム** の階層を指定すると、*operand1* で指定したテキストブロックを見つけるために各階層が順番にスキャンされます。

現在想定されているユーザー ID でアクセスできないキャビネットにドキュメントを保存する場合は、パスワードを指定する必要があります。

`Con-form` は `Con-nect` アクセス制限の準拠を強化し、`Con-nect` に定義済みのキャビネット ID だけを受け入れます。

この副節を省略すると、`Con-form` データエリアが処理されます。

**注意:** キャビネット ID と テキストブロック ID は、大文字で指定する必要があります。



**STATUS 副節**

```
[STATUS operand1 [operand2 [operand3 [operand4]]]]
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	S	A	可	不可
<i>operand2</i>	S	B	可	不可
<i>operand3</i>	S	B	可	不可
<i>operand4</i>	S	B	可	不可

構文要素の説明：

<i>operand 1</i>	Status 変数 「State」 を含みます。
<i>operand 2</i>	Status 変数 「Position (ページ番号)」 を含みます。
<i>operand 3</i>	Status 変数 「Position (行番号)」 を含みます。
<i>operand 4</i>	Status 変数 「Amount of Output Data」 を含みます。

**PROFILE 副節**

```
PROFILE operand1
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S	A	可	不可

この副節を使用すると、指定したテキストブロックの内容が、*INPUT-subclause* で指定されている入力よりも先に処理されます（デフォルトでは、テキストブロックがプロファイルとして処理されることはありません）。

**MESSAGES 副節**


```
MESSAGES { [LISTED] [ON] (rep) }
          { SUPPRESSED }
```

警告メッセージと統計情報は、フォーマットが完了すると表示されます。SUPPRESSED を指定すると、メッセージは表示されず、またエラーは無視されます。

**ERRORS 副節**

```
ERRORS { [LISTED] [ON] (rep) }
        { INTERCEPTED }
```

この副節は、フォーマットエラーが発生した場合に実行されるアクションを指定します。エラーは単に無視されるだけですが、Naturalの標準エラー処理ルーチンによって処理されるか、または Natural レポート (rep) に出力されます。

 **注意:** エラーとメッセージは相互に排他的です。異なるオプションを指定していた場合でも、一部のエラーでは Natural の標準エラー処理ルーチンが呼び出されます。DEFINE PRINTER (n) OUTPUT 'CONNECT' ステートメントによって出力先が Con-nect システムになっているレポートを、エラーまたはメッセージの出力先に指定することはできません。

**ENDING 副節**

```
ENDING { [AT] [PAGE] operand1 }
        { AFTER operand1 [PAGES] }
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
operand1	C S	N P	可	不可

この副節を使用すると、指定した番号の後にあるページでのフォーマット済みテキストの出力を抑制したり、フォーマットされた出力の量を指定したページ数に制限できます。

## STARTING 副節

```
STARTING [FROM] [PAGE] operand1
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S	N P	可	不可

この副節を使用すると、指定した番号 (*operand1*) のページに到達するまで、フォーマットされたテキストの出力が抑制されます。

## EXTRACTING 節

```
EXTRACTING [TEXTVARIABLE] {operand=operand2},... 19
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	S	A N P	可	可
<i>operand2</i>	C S	A	可	不可

この節は、テキスト変数の値を Natural 変数に割り当てます。テキスト変数の現在の設定は、前のフォーマット操作の結果によるものです。

テキスト変数名は、大文字で指定する必要があります。

## フォーマット処理

COMPOSE ステートメントの **FORMATTING** 節が実行されると、フォーマット処理が開始されます（**MOVING** 節によるテキスト入力を行おうとしたが、まだその入力が行われていない場合でも開始されます）。フォーマット処理がアクティブな間は、COMPOSE MOVING ステートメントの実行によって生じるテキスト入力は、フォーマッタのワークスペースに直接送られます（この入力は、後で別のフォーマットプロセスに再使用することはできません）。フォーマット処理が非アクティブな場合、テキスト入力は中間にある「DATAAREA」の COMPOSE バッファに格納されます。したがって、入力を複数のフォーマット処理で再使用できます。

フォーマッタのバッファは Natural プログラムの最後ではクリアされないため、個々の COMPOSE ステートメントを1つの Natural プログラムの中で実行する必要はありません。これらのステートメントは、連続して呼び出される複数のプログラムの中で発行できます。

RESETTING 節や FORMATTING 節が実行されるか、または重大なフォーマットエラーが発生すると、進行中のフォーマット処理は終了します。

入力終了 (EOI) は、MOVING 節の LAST 副節で指定します。

入力ソースとして Con-nect ドキュメントを指定した場合は、ドキュメントの終点に到達すると、そこが入力終了であるとみなされます。



**注意:** フォーマット処理が所定の処理ステップにおいて常に適切なステータスであることを、FORMATTING または MOVING 節の STATUS 副節を使用して確認することをお勧めします。

## ダイアログモード

ダイアログモード処理は、入力のフォーマット中および出力の生成中に、ユーザープログラムとフォーマッタの間で実行される一連の対話です。

ダイアログモードを使用すると、ユーザープログラムで、入力階層の任意のレベルにあるフォーマッタに未加工テキストを入力として提供できます。また、このモードは、フォーマットされた出力を現在のプログラム環境に直接受け入れます。

対話を行うには、フォーマット処理を一連のステップに再分割します。各ステップは、COMPOSE ステートメントによって別々に呼び出されます。

- 入力用のダイアログモード
- 出力用のダイアログモード
- 入出力用のダイアログモード
- ダイアログモードでの COMPOSE ステートメントの実行

### 入力用のダイアログモード

入力テキストのソースが DATAAREA であるか、またはフォーマット制御ステートメント .TE ON に到達したときに、Con-form のデータエリアに処理するデータがそれ以上ない場合は、入力用のダイアログモードに入ります。入力用のダイアログモードは、最初の STATUS 変数にあるワード TERM によって通知されます。

ユーザープログラムでは、その後処理される COMPOSE ステートメントの MOVING 関数を呼び出し、要求された入力を提供することによって、応答する必要があります。MOVING 節の LAST オプションを指定するか、または MOVING 関数からのテキストと同様に .TE ON によって端末入力呼び出された場合は .TE OFF を指定することによって、ユーザープログラムは端末入力を終了できます。フォーマッタでは、フォーマット処理の終了を END または ENDX (最初のステータス変数にエラーがある場合) を使用して通知します。

## 出力用のダイアログモード

出力先を `TO VARIABLES` に指定すると、出力用のダイアログモードに入ります。提供された `Natural` 変数に値が挿入されるか、または改ページに到達すると（いずれか先に発生した方）、制御がフォーマッタから `Natural` プログラムに戻されます。出力用のダイアログモードは、最初の `STATUS` 変数にあるワード `STRG` によって通知されます。ユーザープログラムは、`Natural` 変数に挿入されたフォーマット済み出力を取得して、その後処理される `COMPOSE MOVING` ステートメントで `Natural` 変数の別のセットを出力先に指定することによって、応答する必要があります。フォーマット処理の終了は、`END` または `ENDX`（最初のステータス変数にエラーがある場合）で示されます。

 **注意:** ダイアログモードを使用する場合（「`INPUT` 副節」と「`OUTPUT` 副節」を参照）、通常はフォーマット操作を複数の `COMPOSE` ステートメントに分散して実行します。

## 入出力用のダイアログモード

入力と出力処理を組み合わせたダイアログモードに入ることができます。したがって、フォーマッタによって追加入力が要求されるか（`TERM` で示される）、またはフォーマッタによって出力が提供されると（`STRG` で示される）、`Natural` プログラムでは適切なアクションを実行する必要があります。

入力と出力処理を組み合わせたダイアログモードに入ると、フォーマッタで一度に受け入れられる入力は1行だけになります。入力モードのみの場合は、一度に複数の行が受け入れられます。

## ダイアログモードでの `COMPOSE` ステートメントの実行


一連の `COMPOSE MOVING` の実行を含む `COMPOSE FORMATTING` ステートメントによってダイアログモードに入ったことが示される場合は、次のことに注意してください。

- `COMPOSE ASSIGNING` と `COMPOSE EXTRACTING` ステートメントは、ダイアログモードがアクティブな間は有効です。
- `COMPOSE RESETTING` と `FORMATTING` は、すべてのフォーマット処理を直ちに強制終了します。

## Natural 以外のプログラム

`FORMATTING` 節で指定したパラメータに応じて、入力と出力は `Natural` 以外のプログラムで処理されます。このようなプログラムは、`CALL` ステートメント内で使用されているのと同じメカニズムによって呼び出されます。

`COMPOSE` は、標準のリンケージ規則を使用してこれらのプログラムとパラメータを交換します（`CICS` 環境ではダイナミックロードは不可）。

 **注意:** Natural以外のプログラムによる入力または出力処理は、メインフレームコンピュータ上でのみ可能です。他のプラットフォームでは COMPOSE ステートメントの該当する部分は無視されます。

フォーマット処理のステータスに応じて、フォーマッタと Natural以外のプログラムの間で2つまたは3つのパラメータの受け渡しが行われます。

パラメータ 1 (フォーマット/長さ A1)	ファンクションコードが、フォーマッタから Natural以外のプログラムに渡されます。設定可能値：	
	I	開始（入力、出力）。
	O	ドキュメントをオープン（入力）。
	R	ドキュメントの1行を読み取り（入力）。
	W	出力の1行に書き込み（出力）。
	C	ドキュメントをクローズ（入力）。
	T	終了（入力、出力）。
パラメータ 2 (フォーマット/長さ B1)	レスポンスコードが Natural以外のプログラムからフォーマッタに渡されます。設定可能値：	
	X'00'	機能が正常に完了しました。
	X'01'	機能Oへの応答：ドキュメントが見つかりませんでした。
		機能Rへの応答：ドキュメントの終点に到達しました。
X'FF'	機能が完了していません。	
パラメータ 3 (フォーマット A1/256)	機能OとWの場合、これらのパラメータはフォーマッタから Natural以外のプログラムに渡されます。ただし、機能Rからのパラメータは、Natural以外のプログラムからフォーマッタに渡されます。	
	バイト 1~2	このパラメータの長さ n を表します。
	バイト 3~4	空きです。
	バイト 5~n	機能O：ドキュメント名。
		機能R：Natural以外のプログラムによって読み取られる行。
		機能W：フォーマッタからの出力の行。
	フォームフィードが必要な場合は出力の先頭に N が付けられ、それ以外の場合は 1 が付けられます。	
出力を Natural以外のプログラムに渡す場合は、テキストを強調表示する太字や斜体などの特定のオプションは無視されます。		

## 例

- 例 1
- 例 2
- 例 3
- 例 4
- 例 5

### 例 1

```
COMPOSE RESETTING ALL
      FORMATTING INPUT 'TEXT' FROM CABINET 'TLIB'
      OUTPUT (1)
      MESSAGES LISTED ON (0)
```

上記の COMPOSE ステートメントでは、レポート 1 上に生成される Con-nect キャビネット TLIB 内のテキストブロック TEXT をフォーマットして出力します。エラーおよび統計メッセージは、レポート 0 (デフォルトプリンタ) に表示されます。

### 例 2

```
COMPOSE RESETTING ALL
COMPOSE MOVING '.FI ON' 'This is an example'
COMPOSE MOVING 'for use of Con-form from'
      'within Natural applications' LAST
COMPOSE FORMATTING
```

上記の COMPOSE ステートメントは、レポート 0 (デフォルトプリンタ) にフォーマットされたテキストを出力します。

### 例 3

```
COMPOSE ASSIGNING 'VAR1' = 'Text1', 'VAR2' = 540
```

上記の COMPOSE ステートメントは、Con-nect プロシージャの Con-form テキスト変数 &VAR1 と &VAR2 に値を割り当てます。

**例 4**

XYLIB 内のテキストブロック XYZ :

```
.FI ON
Dear Mr &name.,
.IL
I am pleased to invite you to a presentation of our new product &prod..
```

Natural プログラム :

```
...
INPUT #NAME (A32) #PROD (A32)
COMPOSE ASSIGNING 'NAME' = #NAME, 'PROD' = #PROD
          FORMATTING INPUT 'XYZ' FROM CABINET 'XYLIB'
          OUTPUT (1) MESSAGES SUPPRESSED
...
```

プログラムによって生成された入力マップ :

```
#NAME Davenport
#PROD Natural 4.2
```

出力結果 :

```
Dear Mr Davenport,

I am pleased to invite you to a presentation of our new product Natural 4.2.
```

**例 5**

この例は、入力と出力処理を組み合わせたダイアログモードでのフォーマット処理を示しています。このプログラム例では、Con-form の行単位のフォーマットモードを開始して、一部のコマンドと変数を Con-form に渡し、ステータス情報とフォーマットされた出力行を画面に表示するサブルーチンを実行します。

```
DEFINE DATA LOCAL
01 #LINES_PER_PERFORM(P5) /* counts repeat-loops per PERFORM CNF_OUT
01 #TRACE(A1) INIT<'N'> /* if 'Y' displays additional trace-infos
01 #OUT_FORM(A1) INIT<'F'> /* output-format
01 #START_PAGE (P3) INIT<1> /* beginning of display
01 #CNTR (P5) /* Loop-Counter
01 #STATI /* Status-Information
 02 #STATUS (A4) /* can be STRG TERM END or ENDX
 02 #PAGE (B4) /* actual page-number
 02 #LINE (B4) /* actual line-number on page (not .tt/.bt)
 02 #NO_LINES (B4) /* number of lines returned
```



```

02 REDEFINE #NO_LINES
03 #NO_LINES_I (I4)
01 #OUTPUT(A30/4)          /* output of formatted line
01 #INDEX (P3)            /* index as pointer to out line
END-DEFINE
*
SET KEY ALL
SET CONTROL 'M9'
INPUT
008/008 'Demonstration of formatted output to Variable'(I)
/ 08X 'Enter page to start display      :' #START_PAGE(AD=MIL)
/ 08X 'Display additional trace-data ?:' #TRACE(AD=MIT)
/ 08X 'Please select the output-format:' #OUT_FORM(AD=MIT)
      '(F=Final without BP/US-marks'
/ 44X 'M=Final with BP/US-marks "<>"'
/ 09X '(only, if CMF-Zap 2056 applied =>) I=Intermediate)'
/ 50X 'PF3=Exit'(I)
*
IF *PF-KEY EQ 'PF3'
SET CONTROL 'MB'
STOP
END-IF
*
IF NOT (#OUT_FORM EQ 'F' OR EQ 'M' OR EQ 'I')
REINPUT ' Please enter valid code!' MARK *#OUT_FORM ALARM
END-IF
*
WRITE TITLE LEFT
      'Stat * Page      * Line      * No.Lines >> Formatted Output'(I)
/ '-'(79)(I)
*
SET CONTROL 'MB'
COMPOSE RESETTING ALL /* clear all con-form buffers
RESET #NO_LINES
*
* start line-oriented formatting-mode here
* starting from 0
DECIDE ON FIRST VALUE OF #OUT_FORM
VALUE 'F'
COMPOSE FORMATTING
      OUTPUT TO VARIABLES #OUTPUT (1:4)          /* to Output
      STATUS #STATUS #PAGE #LINE #NO_LINES      /* get Status
VALUE 'M'
COMPOSE FORMATTING
      OUTPUT TO VARIABLES CONTROL '<' '>'
      #OUTPUT (1:4)          /* to output
      STATUS #STATUS #PAGE #LINE #NO_LINES      /* get Status
VALUE 'I'
COMPOSE FORMATTING
      OUTPUT TO VARIABLES CONTROL 'I' 'N'
      #OUTPUT (1:4)          /* to output
      STATUS #STATUS #PAGE #LINE #NO_LINES      /* get Status

```

```

NONE
  STOP
END-DECIDE
*
RESET #NO_LINES
*
* put some commands to con-form to see something
*
COMPOSE MOVING
  '.pl 16;.hs 2;.tt 1Formatierung in Variable//;.tt 2//' /* Cmd
  OUTPUT TO VARIABLES #OUTPUT (1:4) /* to Output
  STATUS #STATUS #PAGE #LINE #NO_LINES /* get Status
*
COMPOSE MOVING
  '.fs 1;.bt Ende Seite #//;.fi on;.tb *=15' /* Commands
  OUTPUT TO VARIABLES #OUTPUT (1:4) /* to Output
  STATUS #STATUS #PAGE #LINE #NO_LINES /* get Status
*
*
* loop 40-times, send commands to con-form and display output
*
COMPOSE ASSIGNING 'Wert' = '1-20' /* Assign value to variable &Wert
*
FOR #CNTR 1 40 /* Loop some time
  IF #STATUS NE 'TERM' /* no wait-for-input => error!!!!
  IF #STATUS EQ 'STRG'
    IGNORE
  ELSE
    WRITE 'Unexpected Status-code' #STATUS(AD=0I) 'found!'
    / 'Execution has stopped....'
    STOP
  END-IF
END-IF
*
IF #CNTR EQ 21
  COMPOSE ASSIGNING 'Wert' = '21-40' /* Assign a variable-value
END-IF
COMPOSE ASSIGNING 'CNTR' = #CNTR /* Again assignment
COMPOSE MOVING
  '.BP;&Wert *Durchlauf &CNTR;.BR' /* Commands
  OUTPUT TO VARIABLES #OUTPUT (1:4) /* to Output
  STATUS #STATUS #PAGE #LINE #NO_LINES /* get Status
  PERFORM CNF-OUT /* show result
END-FOR
COMPOSE MOVING
  LAST /* End-Of-Processing
  OUTPUT TO VARIABLES #OUTPUT (1:4) /* to Output
  STATUS #STATUS #PAGE #LINE #NO_LINES /* get Status
*
IF #TRACE EQ 'Y'
  WRITE 'End of processing...'(I)
END-IF

```

```

*
* Subroutines
*
PERFORM CNF-OUT
*
* Subroutine to display any waiting output from con-form
*
DEFINE SUBROUTINE CNF-OUT
  RESET #LINES_PER_PERFORM
  REPEAT UNTIL #STATUS EQ 'TERM' /* TERM = input waiting
    PERFORM BREAK /* do some break-processing
  AT BREAK OF #PAGE
    IF #PAGE GT #START_PAGE
      WRITE '-'(79)(I)
    END-IF
    IF #TRACE EQ 'Y'
      WRITE 'End of this page...'(I)
    END-IF
    NEWPAGE
  END-BREAK
  IF #PAGE GE #START_PAGE /* show line of output
    IF #NO_LINES_I GT 0
      FOR #INDEX 1 #NO_LINES_I
        ADD 1 TO #LINES_PER_PERFORM /* count loops
        WRITE NOTIT NOHDR #STATUS '*' #PAGE '*' #LINE
          '*' #NO_LINES
          '>>' #OUTPUT (#INDEX)
      END-FOR
    END-IF
  END-IF
  IF #STATUS NE 'STRG' /* if no wait on out
    ESCAPE BOTTOM
  END-IF
  RESET #NO_LINES
  COMPOSE MOVING
    OUTPUT TO VARIABLES #OUTPUT (1:4) /* get Output
    STATUS #STATUS #PAGE #LINE #NO_LINES /* Status
  END-REPEAT
*
  IF #TRACE EQ 'Y'
    WRITE 'Count of Lines per PERFORM was'(I) #LINES_PER_PERFORM(AD=0I)
  END-IF
*
END-SUBROUTINE
SET CONTROL 'MB'
END

```

---

# 24 COMPRESS

---

■ 機能 .....	156
■ 構文説明 .....	156
■ 処理 .....	160
■ 例 .....	160

```

COMPRESS [NUMERIC] [FULL]

{ operand1 [(parameter)]
  SUBSTRING (operand1, operand3, operand4) [(parameter)] }...

INTO { operand2
      SUBSTRING
      (operand2, operand5, operand6) }

[ LEAVING SPACE
  LEAVING NO [SPACE]
  WITH [ALL] [DELIMITERS] [operand7] ]
    
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ASSIGN](#) | [COMPUTE](#) | [EXAMINE](#) | [MOVE](#) | [MOVE ALL](#) | [SEPARATE](#)

関連機能グループ：「[算術演算とデータ移動操作](#)」

## 機能

COMPRESS ステートメントは、1つ以上のオペランドの内容を単一の英数字フィールドへ転送（結合）するために使用します。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
operand1	C S A G N	A U N P I F B D T   G O	可	不可
operand2	S	A U   B	可	可
operand3	C S	N P I B*	可	不可
operand4	C S	N P I B*	可	不可
operand5	C S	N P I B*	可	不可
operand6	C S	N P I B*	可	不可
operand7	C S	A U   B	可	不可

\* *operand3* のフォーマット B、*operand4*、*operand5*、および *operand6* は、4 以下の長さでのみ使用できます。

構文要素の説明：

<b>NUMERIC</b>	<p>このオプションは、符号文字や小数文字の扱いを指定します。</p> <p>NUMERIC を指定しないと、数値の小数点や符号は値が転送される前に除外されます。次に例を示します。</p> <pre>COMPRESS -123 1.23 INTO #TARGET WITH DELIMITER '*' Content of #TARGET is: 123*123</pre> <p>NUMERIC を指定すると、数値の小数点や符号もターゲットフィールドに転送されます。</p> <p>浮動小数点の値については、NUMERIC が指定されているかどうかに関係なく、小数点と符号が転送されます。</p> <p>例 1：</p> <pre>COMPRESS NUMERIC -123 1.23 INTO #TARGET WITH DELIMITER '*' Content of #TARGET is: -123*1.23</pre> <p>例 2：</p> <pre>COMPRESS NUMERIC 'ABC' -0056.00 -0056.10 -0056.01 INTO #TARGET WITH DELIMITER '*' Content of #TARGET is: ABC*-56*-56.1*-56.01</pre> <p>例 3：</p> <pre>COMPRESS NUMERIC FULL 'ABC' -0056.00 -0056.10 -0056.01 INTO #TARGET WITH DELIMITER '*' Content of #TARGET is: ABC*-0056.00*-0056.10*-0056.01</pre>
<b>FULL</b>	<p>FULL を指定しないと、値が転送される前に次のものがソースフィールドから削除されます。</p> <ul style="list-style-type: none"> <li>■ フォーマット N、P、または I のフィールドの小数点の前にある先頭のゼロ</li> <li>■ フォーマット N または P のフィールドの小数点の後にある末尾のゼロ</li> <li>■ フォーマット A のフィールドの末尾にある空白</li> <li>■ フォーマット B のフィールドの先頭にあるバイナリの 0</li> </ul>

	<p>すべて0の数値ソースフィールドについては1つの0が転送されます。次に例を示します。</p> <pre>COMPRESS 'ABC ' 001 INTO #TARGET WITH DELIMITER '*' Content of #TARGET is: ABC*1</pre> <p>FULL を指定すると、ソースフィールドの実際の長さの値がターゲットフィールドに送られます。つまり、次のようになります。</p> <ul style="list-style-type: none"> <li>■ フォーマット N、P、または I のフィールドの小数点の前にある先頭のゼロ</li> <li>■ フォーマット N または P のフィールドの小数点の後にある末尾のゼロ</li> <li>■ フォーマット A のフィールドの末尾にある空白</li> <li>■ フォーマット B のフィールドの先頭にあるバイナリの 0</li> </ul> <p>これらは入力したとおりに表示されます。次に例を示します。</p> <pre>COMPRESS FULL 'ABC ' 001 INTO #TARGET WITH DELIMITER '*' Content of #TARGET is: ABC *001</pre>
<i>operand1</i>	<p>ソースフィールド：</p> <p><i>operand1</i> では、内容が転送されるフィールドを指定します。</p> <p><b>注意:</b> <i>operand1</i> がフォーマット A または B でない場合、その内容は転送される前に英数字表現に変換されます。必要に応じて、英数字表現は切り捨てられます。</p> <p><i>operand1</i> が時刻変数（フォーマット T）の場合は、変数内容のうち時刻コンポーネントのみが転送され、日付コンポーネントは転送されません。</p>
<i>operand2</i>	<p>ターゲットフィールド：</p> <p><i>operand2</i> では、ソースフィールドの値を受け取るフィールドを指定します。</p> <p>ターゲットフィールドがフォーマット U（Unicode）であり、フォーマット B のソースフィールドが含まれている場合、送信するバイナリフィールドの長さを同じにする必要があります。</p>
<b>LEAVING SPACE</b>	<p>追加オプションを一切指定しないで COMPRESS ステートメントを使用したか、または LEAVING SPACE（デフォルトでも適用される）を指定した場合、ターゲットフィールド内の値は互いに空白で区切られます。</p>
<b>LEAVING NO SPACE</b>	<p>LEAVING NO SPACE が指定されると、ターゲットフィールド内の各値の間には、空白も文字も入りません。</p>
<i>parameter</i>	<p><i>parameter</i> では、セッションパラメータ PM またはセッションパラメータ DF を指定できません。</p>



	PM=I	<p>右から左に記述する言語をサポートするために、PM=I を指定して、<i>operand1</i> の値を逆（右から左）方向で <i>operand2</i> に渡すことができます。例えば、次のステートメントの結果として、#B の内容は ZYXABC になります。</p> <pre>MOVE 'XYZ' TO #A COMPRESS #A (PM=I) 'ABC' INTO #B LEAVING NO SPACE</pre> <p><i>operand1</i> の末尾にある空白はすべて削除され（FULL を指定した場合を除く）、値は1文字ずつ逆向きに並べ換えられて <i>operand2</i> に転送されます。</p>
	DF	<p><i>operand1</i> が日付変数の場合、変数に対する <i>parameter</i> としてセッションパラメータ DF を指定できます。</p>
<b>SUBSTRING</b> <i>(operand1,</i> <i>operand3,</i> <i>operand4)</i>		<p><i>operand1</i> が英数字 (A)、Unicode (U)、またはバイナリ (B) フォーマットの場合、SUBSTRING オプションを使用して、ソースフィールドの特定部分だけを転送できます。フィールド名 (<i>operand1</i>) の後に、まず開始位置 (<i>operand3</i>) を指定してから、フィールドの転送される部分の長さ (<i>operand4</i>) を指定してください。</p>
<b>INTO SUBSTRING</b> <i>(operand2,</i> <i>operand5,</i> <i>operand6)</i>		<p>INTO 節に SUBSTRING オプションを使用してソース値をターゲットフィールドの特定部分に転送することもできます。</p> <p>どちらの場合も、COMPRESS ステートメントの SUBSTRING オプションの使用は、MOVE ステートメントの SUBSTRING オプションに対応しています。SUBSTRING オプションの詳細については、<a href="#">MOVE ステートメント</a>を参照してください。</p>
<b>WITH DELIMITERS</b>		<p>DELIMITERS オプションは、ターゲットフィールドの値を特定の文字で区切ることができます。</p> <p>WITH DELIMITERS を <i>operand7</i> なしで指定した場合、値は INPUT 区切り文字（セッションパラメータ ID で定義した値）で区切られます。</p>
<b>WITH DELIMITERS</b> <i>operand7</i>		<p>WITH DELIMITERS <i>operand7</i> を指定すると、値は <i>operand7</i> に指定した文字で区切られます。<i>operand7</i> は、1文字にする必要があります。<i>operand7</i> が変数の場合は、そのフォーマット/長さを (A1) または (B1) にする必要があります</p> <p>ターゲットフィールドがフォーマット A または B の場合は、デリミタ文字のフォーマット/長さを (A1)、(B1)、または (U1) にする必要があります。</p> <p>ターゲットフィールドがフォーマット U (Unicode) の場合は、デリミタ文字のフォーマット/長さを (A1)、(B2)、または (U1) にする必要があります。</p>

<b>WITH ALL</b>	<p>ALL を指定しないと、ターゲットフィールドでは、実際に転送された値の間にのみデリミタ文字が挿入されます。次に例を示します。</p> <pre>COMPRESS 'A' ' ' 'C' ' ' INTO #TARGET WITH DELIMITERS '*' Content of #TARGET is: A*C</pre> <p>ALL を指定すると、ターゲットフィールドでは、実際には転送されない空値に対してもデリミタ文字が挿入されます。デリミタ文字の個数は、ソースフィールド数から1を引いた数になります。これは、ターゲットフィールドの内容を後続の <b>SEPARATE</b> ステートメントで再度分割するときなどに有用です。次に例を示します。</p> <pre>COMPRESS 'A' ' ' 'C' ' ' INTO #TARGET WITH ALL DELIMITERS '*' Content of #TARGET is: A**C*</pre>
-----------------	---

## 処理

フォーマット B の転送先フィールドは、フォーマット A の転送先フィールドと同様に処理されます。

COMPRESS 操作は、全オペランドの処理が終了するか、またはターゲットフィールド (*operand2*) に空きがなくなると終了します。

ターゲットフィールドが全オペランドの合計桁数よりも大きい場合、*operand2* の残りの部分は空白となります。ターゲットフィールドの方が小さい場合、値は切り捨てられます。

*operand2* が DYNAMIC 変数である場合、COMPRESS 操作は全ソースオペランドが処理されると終了します。値の切り捨ては行われません。COMPRESS 操作後の *operand2* の長さは、ソースオペランドの合計の長さに対応します。ダイナミック変数の現在の長さは、システム変数 \*LENGTH を使用して確認できます。

## 例

このセクションでは、次のトピックについて説明します。

- [例 1 - COMPRESS](#)
- [例 2 - COMPRESS LEAVING NO SPACE](#)

### ■ 例 3 - COMPRESS WITH DELIMITERS

#### 例 1 - COMPRESS

```

** Example 'CMPEX1': COMPRESS
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 MIDDLE-I
*
1 #COMPRESSED-NAME (A20)
END-DEFINE
*
LIMIT 4
READ EMPLOY-VIEW BY NAME
  COMPRESS FIRST-NAME MIDDLE-I NAME INTO #COMPRESSED-NAME
  DISPLAY NOTITLE
    FIRST-NAME MIDDLE-I NAME 5X #COMPRESSED-NAME
END-READ
*
END

```

プログラム **CMPEX1** の出力：

FIRST-NAME	MIDDLE-I	NAME	#COMPRESSED-NAME
KEPA		ABELLAN	KEPA ABELLAN
ROBERT	W	ACHIESON	ROBERT W ACHIESON
SIMONE		ADAM	SIMONE ADAM
JEFF	H	ADKINSON	JEFF H ADKINSON

#### 例 2 - COMPRESS LEAVING NO SPACE

```

** Example 'CMPEX2': COMPRESS (with LEAVING NO SPACE)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CURR-CODE (1)
  2 SALARY (1)
*
1 #CCSALARY (A20)
END-DEFINE
*
LIMIT 4
READ EMPL-VIEW BY NAME

```

```

COMPRESS CURR-CODE (1) SALARY (1) INTO #CCSALARY
      LEAVING NO SPACE
DISPLAY NOTITLE
      NAME CURR-CODE (1) SALARY (1) 5X #CCSALARY
END-READ
*
END

```

プログラム **CMPEX2** の出力：

NAME	CURRENCY CODE	ANNUAL SALARY	#CCSALARY
ABELLAN	PTA	1450000	PTA1450000
ACHIESON	UKL	11300	UKL11300
ADAM	FRA	159980	FRA159980
ADKINSON	USD	34500	USD34500

### 例 3 - COMPRESS WITH DELIMITERS

```

** Example 'CMPEX3': COMPRESS (with delimiter)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CURR-CODE (1)
  2 SALARY (1)
*
1 #CCSALARY (A20)
END-DEFINE
*
LIMIT 4
READ EMPL-VIEW BY NAME
  COMPRESS CURR-CODE (1) SALARY (1) INTO #CCSALARY
    WITH DELIMITER '*'
  DISPLAY NOTITLE NAME CURR-CODE (1) SALARY (1) 5X #CCSALARY
END-READ
*
END

```

プログラム **CMPEX3** の出力：

NAME	CURRENCY CODE	ANNUAL SALARY	#CCSALARY
ABELLAN	PTA	1450000	PTA*1450000
ACHIESON	UKL	11300	UKL*11300
ADAM	FRA	159980	FRA*159980
ADKINSON	USD	34500	USD*34500



# 25 COMPUTE

---

▪ 機能 .....	166
▪ 構文説明 .....	168
▪ 除算結果の精度 .....	170
▪ SUBSTRING オプション .....	170
▪ 例 .....	170

ストラクチャードモード構文

```

{
  { COMPUTE } [ROUNDED] { operand1 [:=] ... { arithmetic-expression }
  { ASSIGN } { operand2 }
}
{ operand1 := } ... { arithmetic-expression }
{ operand2 }

```

レポートニングモード構文

```

[ { COMPUTE } ] [ROUNDED] { operand1[:=] { arithmetic-expression }
[ { ASSIGN } ] { ... } { operand2 }

```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：**ADD** | **COMPRESS** | **DIVIDE** | **EXAMINE** | **MOVE** | **MOVE ALL** | **MULTIPLY** | **RESET** | **SEPARATE** | **SUBTRACT**

関連機能グループ：「[算術演算とデータ移動操作](#)」

## 機能

COMPUTE ステートメントは、算術演算や割り当て操作で使用します。

COMPUTE ステートメントで複数のターゲットオペランド (*operand1*) を指定しても、ソースオペランド (*operand2*) が算術演算式はない場合は、個別の COMPUTE ステートメントと同じになります。

```
#TARGET1 := #TARGET2 := #SOURCE
```

これは、以下と同じです。

```
#TARGET1 := #SOURCE
#TARGET2 := #SOURCE
```



例：

```

DEFINE DATA LOCAL
1 #ARRAY(I4/1:3) INIT <3,0,9>
1 #INDEX(I4)
1 #RESULT(I4)
END-DEFINE
*
#INDEX := 1
*
#INDEX :=      /* #INDEX is 3
#RESULT :=     /* #RESULT is 9
#ARRAY(#INDEX)
*
#INDEX := 2
*
#INDEX :=      /* #INDEX is 0
#ARRAY(3) :=   /* returns run time error NAT1316
#ARRAY(#INDEX)
END

```

ソースオペランドが算術演算式の場合は、式が評価され、その結果は一時変数に格納されます。その後、一時変数はターゲットオペランドに割り当てられます。

```

#TARGET1 := #TARGET2 := #SOURCE1 + 1
is identical to
#TEMP := #SOURCE1 + 1
#TARGET1 := #TEMP
#TARGET2 := #TEMP

```

例：

```

DEFINE DATA LOCAL
1 #ARRAY(I4/1:3) INIT <2, 0, 9>
1 #INDEX(I4)
1 #RESULT(I4)
END-DEFINE
*
#INDEX := 1
*
#INDEX :=      /* #INDEX is 3
#RESULT :=     /* #RESULT is 3
#ARRAY(#INDEX) + 1
*
#INDEX := 2
*
#INDEX :=      /* #INDEX is 0
#ARRAY(3) :=   /* returns run time error NAT1316
#ARRAY(#INDEX)
END

```

詳細については、『プログラミングガイド』の「演算割り当てのルール」、および特に次のセクションを参照して下さい。

- 配列での算術演算
- 「データ転送」（データ転送の互換性およびデータ転送のルールに関する情報）

## 構文説明

オペランド定義テーブル：

オペランド	構文要素			フォーマット										オペランド参照	ダイナミック定義					
<i>operand1</i>	S	A	M	A	U	N	P	I	F	B	D	T	L	C	G	O	可	可		
<i>operand2</i>	C	S	A	N	E	A	U	N	P	I	F	B	D	T	L	C	G	O	可	不可

構文要素の説明：

<b>COMPUTE ASSIGN [:]=</b>	<p>このステートメントは、ステートメントキーワード COMPUTE（または ASSIGN）を省略して短縮形で発行することができます。</p> <p>ストラクチャードモードでステートメントキーワード COMPUTE（または ASSIGN）を省略する場合には、割り当て文字 (=) の前にコロン (:) を指定します。</p> <p>ただし、ROUNDED オプションを使用する場合は、ステートメント名である COMPUTE または ASSIGN を指定する必要があります。</p>
<b>ROUNDED</b>	<p>キーワード ROUNDED を指定すると、値は <i>operand1</i> に割り当てられる前に切り上げられます。切り上げの詳細については、『プログラミングガイド』の「演算割り当てのルール」、および「フィールドの切り捨てと切り上げ」を参照してください。</p>
<i>operand1</i>	<p>結果フィールド：</p> <p><i>operand1</i> には、算術／割り当て演算の結果が格納されます。</p> <p>結果の精度については、『プログラミングガイド』の「算術演算結果の精度」を参照してください。</p> <p><i>operand1</i> がデータベースフィールドでも、データベースのそのフィールドは更新されません。</p> <p><i>operand1</i> がダイナミック変数の場合は、<i>operand2</i> の長さまたは算術演算の結果の長さまで値が入り、それに応じて <i>operand1</i> の長さは調整されます。ダイナミック変数の現在の長さは、システム変数 *LENGTH を使用して確認できます。</p>

	<p>ダイナミック変数の全般的な情報については、「<a href="#">レンジ変数／フィールドとダイナミック変数／フィールド</a>」を参照してください。</p>														
<i>arithmetic-expression</i>	<p>算術演算式は、1つ以上の定数、データベースフィールド、およびユーザー定義変数で構成されます。</p> <p>Natural算術関数（『システム関数』の算術関数を参照）は、算術オペランドとしても使用できます。</p> <p>算術演算式のオペランドは、N、P、I、F、D、またはTフォーマットで定義する必要があります。</p> <p>オペランドのフォーマットについては、『プログラミングガイド』の「<a href="#">フォーマット混合式のパフォーマンスについて</a>」を参照してください。</p> <p>次の結合演算子を使用できます。</p> <table border="1"> <thead> <tr> <th>演算子</th> <th>記号</th> </tr> </thead> <tbody> <tr> <td>カッコ</td> <td>()</td> </tr> <tr> <td>累乗</td> <td>**</td> </tr> <tr> <td>乗算</td> <td>*</td> </tr> <tr> <td>除算</td> <td>/</td> </tr> <tr> <td>加算</td> <td>+</td> </tr> <tr> <td>減算</td> <td>-</td> </tr> </tbody> </table> <p>各演算子の前後には少なくとも1つの空白を挿入して、上記の文字を持つ変数名と混同しないようにしてください。</p> <p>算術演算の処理順序は次のとおりです。</p> <ol style="list-style-type: none"> <li>1. カッコ</li> <li>2. 累乗</li> <li>3. 乗算／除算（現れた順に左から右へ）</li> <li>4. 加算／減算（現れた順に左から右へ）</li> </ol>	演算子	記号	カッコ	()	累乗	**	乗算	*	除算	/	加算	+	減算	-
演算子	記号														
カッコ	()														
累乗	**														
乗算	*														
除算	/														
加算	+														
減算	-														
<i>operand2</i>	<p>ソースフィールド：</p> <p><i>operand2</i>はソースフィールドです。<i>operand1</i>がフォーマットCの場合は、<i>operand2</i>を属性定数として指定することもできます（『プログラミングガイド』の「<a href="#">ユーザー定義定数</a>」を参照）。</p>														

## 除算結果の精度

COMPUTE ステートメントの除算結果の精度（桁数）は、1 番目のオペランド（被除数）または最初の結果フィールドのどちらか大きい方の精度により決まります。

ただし、整数オペランドの除算には、次のことが適用されます。2つの整数の除算について、その結果の精度は最初の結果フィールドの精度により決まりますが、2つの整数オペランドの1つでも変数であれば、結果も整数フォーマットになります。つまり、結果フィールドの精度に関わらず小数桁なしになります。

## SUBSTRING オプション

オペランドが英数字、Unicode、またはバイナリフォーマットの場合は、**SUBSTRING** オプションを使用して *operand2* の一部分を *operand1* に割り当てることができます。記述方法は **MOVE** ステートメントと同じです。

## 例

- 例1 - ASSIGN ステートメント
- 例2 - COMPUTE ステートメント

### 例1 - ASSIGN ステートメント

```
** Example 'ASGEX1S': ASSIGN (structured mode)
*****
DEFINE DATA LOCAL
1 #A (N3)
1 #B (A6)
1 #C (N0.3)
1 #D (N0.5)
1 #E (N1.3)
1 #F (N5)
1 #G (A25)
1 #H (A3/1:3)
END-DEFINE
*
ASSIGN #A = 5                               WRITE NOTITLE '=' #A
ASSIGN #B = 'ABC'                           WRITE '=' #B
ASSIGN #C = .45                             WRITE '=' #C
ASSIGN #D = #E = -0.12345                   WRITE '=' #D / '=' #E
ASSIGN ROUNDED #F = 199.999                 WRITE '=' #F
#G      := 'HELLO'                          WRITE '=' #G
```

```
#H (1) := 'UVW'
#H (3) := 'XYZ'          WRITE '=' #H (1:3)
*
END
```

プログラム **ASGEX1S** の出力：

```
#A:    5
#B: ABC
#C:   .450
#D: -.12345
#E: -0.123
#F:   200
#G: HELLO
#H: UVW    XYZ
```

レポートモードの例については、ライブラリ SYSEXRM のプログラム **ASGEX1R** を参照してください。

## 例2 - COMPUTE ステートメント

```
** Example 'CPTX1': COMPUTE
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 SALARY      (1:2)
*
1 #A           (P4)
1 #B           (N3.4)
1 #C           (N3.4)
1 #CUM-SALARY (P10)
1 #I           (P2)
END-DEFINE
*
COMPUTE #A = 3 * 2 + 4 / 2 - 1
WRITE NOTITLE 'COMPUTE #A = 3 * 2 + 4 / 2 - 1' 10X '=' #A
*
COMPUTE ROUNDED #B = 3 -4 / 2 * .89
WRITE 'COMPUTE ROUNDED #B = 3 -4 / 2 * .89' 5X '=' #B
*
COMPUTE #C = SQRT (#B)
WRITE 'COMPUTE #C = SQRT (#B)' 18X '=' #C
*
LIMIT 1
READ EMPLOY-VIEW BY PERSONNEL-ID STARTING FROM '20017000'
  WRITE / 'CURRENT SALARY: ' 4X SALARY (1)
  / 'PREVIOUS SALARY:' 4X SALARY (2)
  FOR #I = 1 TO 2
    COMPUTE #CUM-SALARY = #CUM-SALARY + SALARY (#I)
```

## COMPUTE

---

```
END-FOR  
WRITE 'CUMULATIVE SALARY:' #CUM-SALARY  
END-READ  
*  
END
```

プログラム **CPTEX1** の出力：

```
COMPUTE #A = 3 * 2 + 4 / 2 - 1      #A:      7  
COMPUTE ROUNDED #B = 3 -4 / 2 * .89 #B:     1.2200  
COMPUTE #C = SQRT (#B)             #C:     1.1045  
  
CURRENT SALARY:          34000  
PREVIOUS SALARY:        32300  
CUMULATIVE SALARY:      66300
```

# 26 CREATE OBJECT

---

▪ 機能 .....	174
▪ 構文説明 .....	174

```
CREATE OBJECT operand1 OF [CLASS] operand2
[GIVING operand4]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[DEFINE CLASS](#) | [INTERFACE](#) | [METHOD](#) | [PROPERTY](#) | [SEND METHOD](#)

関連機能グループ：「[コンポーネントベースプログラミング](#)」

## 機能

CREATE OBJECT ステートメントは、クラスのインスタンスを作成します。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	S		不可	不可
<i>operand2</i>	C S	A	可	不可
<i>operand4</i>	S	N       I	可	不可

構文要素の説明：

<i>operand1</i>	<p>オブジェクトハンドル：</p> <p><i>operand1</i>は、オブジェクトハンドル (HANDLE OF OBJECT) として定義する必要があります。オブジェクトが正常に作成されると、オブジェクトハンドルには値が挿入されます。値が正常に返されなかった場合、<i>operand1</i> の値は NULL-HANDLE になります。</p>
OF CLASS <i>operand2</i>	<p>クラス名：</p> <p><i>operand2</i>は、作成するオブジェクトのクラスの名前です。DCOM クラスとして登録されていないクラスについては、<a href="#">DEFINE CLASS</a> ステートメントで定義したクラス名を割り当てる必要があります。DCOM クラスとして登録されている名前については、クラスの ProgID またはクラス GUID のどちらかを割り当てる必要があります。DCOM クラスとして登録される Natural クラスの場合、ProgID は <a href="#">DEFINE CLASS</a> ステートメントに指定されたクラス名に対応しています。</p>



	CREATE OBJECT #01 OF CLASS "Employee" or CREATE OBJECT #01 OF CLASS "653BCFE0-84DA-11D0-BEB3-10005A66D231"
<b>GIVING</b> <i>operand4</i>	<b>GIVING</b> 節： GIVING 節を指定した場合、 <i>operand4</i> の値は、エラー発生時は Natural メッセージ番号、成功時はゼロのいずれかになります。 GIVING 節を指定しない場合は、エラー発生時に Natural ランタイムエラー処理がトリガされます。



# 27 DECIDE FOR

---

■ 機能 .....	178
■ 構文説明 .....	178
■ 例 .....	179

```

DECIDEFOR      { FIRST }      CONDITION
                { EVERY }
{WHEN logical-condition statement ...} ...
[WHEN ANY statement ...]
[WHEN ALL statement ...]
WHEN NONE statement ...
END-DECIDE

```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[DECIDE ON](#) | [IF](#) | [IF SELECTION](#) | [ON ERROR](#)

関連機能グループ：[論理条件の処理](#)

## 機能

DECIDE FOR ステートメントは、複数の条件（ケース）に応じて1つ以上のアクションを実行するかどうかを判定します。



**注意:** ある条件のもとでは何も実行しない場合は、DECIDE FOR ステートメントの対応する節にステートメント [IGNORE](#) を指定する必要があります。

## 構文説明

<b>FIRST CONDITION</b>	最初に真になった条件のみが処理されます。「 <a href="#">例1</a> 」も参照してください。
<b>EVERY CONDITION</b>	真になった条件がすべて処理されます。「 <a href="#">例2</a> 」も参照してください。
<b>WHEN</b> <i>logical-condition statement</i>	この節では、処理される論理条件（複数可）を指定します。『プログラミングガイド』の「 <a href="#">論理条件基準</a> 」を参照してください。
<b>WHEN ANY</b> <i>statement</i>	WHEN ANY では、任意の論理条件が真になったときに実行される <a href="#">ステートメント（複数可）</a> を指定できます。
<b>WHEN ALL</b> <i>statement</i>	WHEN ALL では、すべての論理条件が真になったときに実行される <a href="#">ステートメント（複数可）</a> を指定できます。この節は、 <a href="#">EVERY</a> が指定された場合にのみ適用されます。
<b>WHEN NONE</b> <i>statement</i>	WHEN NONE では、真になっている論理条件が1つもないときに実行される <a href="#">ステートメント（複数可）</a> を指定できます。
<b>END-DECIDE</b>	DECIDE FOR ステートメントを終了するには、Natural 予約語 END-DECIDE を使用する必要があります。

## 例

- 例 1 - FIRST オプションを指定した DECIDE FOR
- 例 2 - EVERY オプションを指定した DECIDE FOR

### 例 1 - FIRST オプションを指定した DECIDE FOR

```

** Example 'DECEX1': DECIDE FOR (with FIRST option)
*****
DEFINE DATA LOCAL
1 #FUNCTION (A1)
1 #PARM      (A1)
END-DEFINE
*
INPUT #FUNCTION #PARM
*
DECIDE FOR FIRST CONDITION
  WHEN #FUNCTION = 'A' AND #PARM = 'X'
    WRITE 'Function A with parameter X selected.'
  WHEN #FUNCTION = 'B' AND #PARM = 'X'
    WRITE 'Function B with parameter X selected.'
  WHEN #FUNCTION = 'C' THRU 'D'
    WRITE 'Function C or D selected.'
  WHEN NONE
    REINPUT 'Please enter a valid function.'
    MARK *#FUNCTION
END-DECIDE
*
END

```

プログラム **DECEX1** の出力：

```
#FUNCTION A #PARM Y
```

**Enter** キーを押した後：

```
PLEASE ENTER A VALID FUNCTION
#FUNCTION A #PARM Y
```

**例 2 - EVERY オプションを指定した DECIDE FOR**

```
** Example 'DECEX2': DECIDE FOR (with EVERY option)
*****
DEFINE DATA LOCAL
1 #FIELD1 (N5.4)
END-DEFINE
*
INPUT #FIELD1
*
DECIDE FOR EVERY CONDITION
  WHEN #FIELD1 >= 0
    WRITE '#FIELD1 is positive or zero.'
  WHEN #FIELD1 <= 0
    WRITE '#FIELD1 is negative or zero.'
  WHEN FRAC(#FIELD1) = 0
    WRITE '#FIELD1 has no decimal digits.'
  WHEN ANY
    WRITE 'Any of the above conditions is true.'
  WHEN ALL
    WRITE '#FIELD1 is zero.'
  WHEN NONE
    IGNORE
END-DECIDE
*
END
```

プログラム **DECEX2** の出力：

```
#FIELD1 42
```

**Enter** キーを押した後：

```
Page      1                                05-01-11  14:56:26

#FIELD1 is positive or zero.
#FIELD1 has no decimal digits.
Any of the above conditions is true.
```

# 28

## DECIDE ON

---

■ 機能 .....	182
■ 構文説明 .....	182
■ 例 .....	183

```

DECIDE ON      { FIRST
                EVERY }      [VALUES] [OF] operand1
{VALUES operand2 [,operand2] ... [:operand2]} statement ...
[ANY[VALUES] statement ...]
[ALL[VALUES] statement ...]
NONE [VALUES] statement ...
END-DECIDE
    
```


構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[DECIDE FOR](#) | [IF](#) | [IF SELECTION](#) | [ON ERROR](#)

関連機能グループ：[「論理条件の処理」](#)

## 機能

DECIDE ON ステートメントでは、変数の値（1つまたは複数の値）に応じて行う複数の処理を指定します。

 **注意:** ある条件のもとでは何も実行しない場合は、DECIDE ON ステートメントの対応する節にステートメント `IGNORE` を指定する必要があります。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	S A N A U N P I F B D T L G O		可	不可
<i>operand2</i>	C S A	A U N P I F B D T L G O	可	不可

\* *operand5* のフォーマット B、*operand6*、*operand7*、および *operand8* は、4 以下の長さでのみ使用できます。



構文要素の説明：

<b>FIRST/EVERY</b>	これらのキーワードの1つで、見つかった最初の値に対してだけ処理を行うか、または見つかったすべての値に対して処理を行うかを指定します。
<i>operand1</i>	選択フィールド：  <i>operand1</i> または <i>operand2</i> では、内容がチェックされるフィールドの名前を指定します。
<b>VALUES</b> <i>operand2</i> [[ <i>operand2</i> ] ... [: <i>operand2</i> ] <i>statement</i> ...	この節では、選択フィールドの値 ( <i>operand2</i> )、およびフィールドがその値になった場合に実行される <i>statement(s)</i> を指定します。  選択フィールドには、1つの値、複数の値、またはオプションで1つ以上の値の範囲を指定できます。  複数の値を指定する場合は、INPUT区切り文字 (IDセッションパラメータで指定した文字) またはコンマ (,) で区切る必要があります。セッションパラメータ DC でコンマを小数点文字として定義している場合は、この目的でコンマを使用することはできません。  値の範囲を指定する場合、範囲の開始値と終了値をコロン (:) で区切って指定します。
<b>ANY</b> <i>statement</i>	ANY では、VALUES 節にある値のいずれかが見つかった場合に実行される <i>statement(s)</i> を指定します。VALUES 節で指定されたステートメントに加えて、これらのステートメントが実行されます。
<b>ALL</b> <i>statement</i>	ALL では、VALUES 節にある値がすべて見つかった場合に実行される <i>statement(s)</i> を指定します。VALUES 節で指定されたステートメントに加えて、これらのステートメントが実行されます。  ALL 節は、キーワード EVERY を指定した場合にのみ適用されます。
<b>NONE</b> <i>statement</i>	NONE では、指定した値が1つも見つからない場合に実行される <i>statement(s)</i> を指定します。
<b>END-DECIDE</b>	DECIDE ON ステートメントを終了するには、Natural 予約語 END-DECIDE を使用する必要があります。

## 例

- 例 1 - FIRST オプションを指定した DECIDE ON

## ■ 例2 - EVERY オプションを指定した DECIDE ON

### 例1 - FIRST オプションを指定した DECIDE ON

```

** Example 'DECEX3': DECIDE ON (with FIRST option)
*****
*
SET KEY ALL
INPUT 'Enter any PF key' /
      'and check result' /
*
DECIDE ON FIRST VALUE OF *PF-KEY
  VALUE 'PF1'
    WRITE 'PF1 key entered.'
  VALUE 'PF2'
    WRITE 'PF2 key entered.'
  ANY VALUE
    WRITE 'PF1 or PF2 key entered.'
  NONE VALUE
    WRITE 'Neither PF1 nor PF2 key entered.'
END-DECIDE
*
END

```

プログラム **DECEX3** の出力：

```

Enter any PF key
and check result

```

**PF1** を押した後の出力：

```

Page      1                               05-01-11  15:08:50

PF1 key entered.
PF1 or PF2 key entered.

```

### 例2 - EVERY オプションを指定した DECIDE ON

```

** Example 'DECEX4': DECIDE ON (with EVERY option)
*****
DEFINE DATA LOCAL
1 #FIELD (N1)
END-DEFINE
*
INPUT 'Enter any value between 1 and 9:' #FIELD (SG=OFF)
*
DECIDE ON EVERY VALUE OF #FIELD
  VALUE 1 : 4

```

```
    WRITE 'Content of #FIELD is 1-4'  
    VALUE 2 : 5  
    WRITE 'Content of #FIELD is 2-5'  
    ANY VALUE  
    WRITE 'Content of #FIELD is 1-5'  
    ALL VALUE  
    WRITE 'Content of #FIELD is 2-4'  
    NONE VALUE  
    WRITE 'Content of #FIELD is not 1-5'  
    END-DECIDE  
*  
END
```

プログラム **DECEX4** の出力：

```
ENTER ANY VALUE BETWEEN 1 AND 9: 4
```

4を入力して確認した後：

```
Page      1                                05-01-11  15:11:45
```

```
Content of #FIELD is 1-4  
Content of #FIELD is 2-5  
Content of #FIELD is 1-5  
Content of #FIELD is 2-4
```



# 29 DEFINE CLASS

---

▪ 機能 .....	188
▪ 構文説明 .....	188

```

DEFINE CLASS class-name
[ OBJECT          { USING { local-data-area } } ] ...
                   { USING { parameter-data-area } } ] ...
                   data-definition ...
[ LOCAL          { USING { local-data-area } } ] ...
                   { USING { parameter-data-area } } ] ...
                   data-definition
[ INTERFACE USING copycode ] ...
[ INTERFACE      ] ...
[PROPERTY] ...
[METHOD] ...
END-CLASS
    
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CREATE OBJECT](#) | [INTERFACE](#) | [METHOD](#) | [PROPERTY](#) | [SEND METHOD](#)

関連機能グループ：「[コンポーネントベースプログラミング](#)」

## 機能

DEFINE CLASS ステートメントは、Natural クラスモジュール内からクラスを指定するために使  
用します。Natural クラスモジュールは、DEFINE CLASS ステートメントとそれに続く END ステ  
ートメントで構成されています。

## 構文説明

<i>class-name</i>	<p>これは、このクラスのオブジェクトを作成するためにクライアントが使用する名前です。名前の長さは、最大32文字までです。名前にはピリオドが含まれていてもかまいません。ピリオドを使用して、クラス名を次のように構成できます。</p> <p><i>company-name. application-name. class-name</i></p> <p>ピリオド (...) 間にある各部分は、それぞれがユーザー定義変数の命名規則に従っている必要があります（「<a href="#">ユーザー定義変数の命名規則</a>」を参照）。</p> <p>異なるプログラミング言語で書かれたクライアントによってクラスを使用する予定であれば、クラス名はこれらの言語に適用する命名規則と矛盾しないように選択する必要があります。</p>
-------------------	---

<b>OBJECT</b>	OBJECT 節は、オブジェクトデータを定義するために使用します。OBJECT 節の構文は、DEFINE DATA ステートメントの LOCAL 節と同じです。詳細については、 <a href="#">DEFINE DATA</a> ステートメントの <a href="#">LOCAL</a> 節に関する説明を参照してください。
<b>LOCAL</b>	LOCAL 節は、グローバルユニーク ID (GUID) をクラス定義に含めるためにのみ使用します。GUID は、クラスが DCOM で登録される場合にのみ定義する必要があります。GUID は、主にローカルデータエリア内で定義されます。  LOCAL 節の構文は、 <a href="#">DEFINE DATA</a> ステートメントの <a href="#">LOCAL</a> 節と同じです。詳細については、 <a href="#">DEFINE DATA</a> ステートメントの LOCAL 節に関する説明を参照してください。
<b>ID</b>	ID 節は、グローバルユニーク ID をクラスに割り当てるために使用します。クラス GUID は、 <a href="#">LOCAL</a> 節に含まれるデータエリア内に定義される GUID の名前です。クラス GUID は、(命名された) 英数字の定数です。クラスが DCOM で登録される場合は、GUID をクラスに割り当てる必要があります。
<b>INTERFACE USING</b>	INTERFACE USING 節は、 <a href="#">INTERFACE</a> ステートメントを含むコピーコードを組み込むために使用します。
<i>copycode</i>	INTERFACE USING 節で使用するコピーコードには、1 つ以上の <a href="#">INTERFACE</a> ステートメントを含めることができます。
<b>PROPERTY</b>	<a href="#">PROPERTY</a> ステートメントは、インターフェイス定義外で、実装としてオブジェクトデータ変数オペランドをプロパティに割り当てます。
<b>METHOD</b>	<a href="#">METHOD</a> ステートメントは、インターフェイス定義外で、サブプログラムをメソッドに割り当てます。
<b>END-CLASS</b>	DEFINE CLASS ステートメントを終了するには、Natural 予約語 END-CLASS を使用する必要があります。

---



# 30 DEFINE DATA

全般的な構文

```
DEFINE DATA
  [GLOBAL USING global-data-area [WITH block[.block] ...]]
  [
    PARAMETER { USING parameter-data-area } ] ...
                { parameter-data-definition... } ]
  [
    OBJECT { USING { local-data-area } } ] ...
              { parameter-data-area } } ]
              { data-definition... } ]
  [
    LOCAL { USING { local-data-area } } ] ...
              { parameter-data-area } } ]
              { data-definition... } ]
  [INDEPENDENT AIV-data-definition ...]
  [
    CONTEXT { USING { local-data-area } } ]
               { parameter-data-area } } ]
               { context-data-definition ... } ]
END-DEFINE
```

DEFINE DATA ステートメントは、ローカルデータエリア (LDA)、グローバルデータエリア (GDA) またはパラメータデータエリア (PDA) に含まれる定義済みのデータ定義を参照するか、インライン定義を書き込むことにより、Natural プログラムで使用するデータ定義を宣言する多数の節を提供します。

DEFINE DATA ステートメントに関するドキュメントは、次のセクションに分かれています。

- [構文の概要](#)
- [DEFINE DATA - 全般](#)

特定のデータ定義：

- [ローカルデータの定義](#)
- [グローバルデータの定義](#)
- [パラメータデータの定義](#)
- [アプリケーションに依存しない変数の定義](#)
- [Natural RPC 用のコンテキスト変数の定義](#)
- [NaturalX オブジェクトの定義](#)

節とオプション：

- [変数定義](#)
- [ビューの定義](#)
- [再定義](#)
- [ハンドルの定義](#)
- [配列の次元の定義](#)
- [初期値の定義](#)
- [配列用の初期値／定数値](#)
- [フィールド／変数の EM、HD、PD パラメータ](#)

例：

- [DEFINE DATA ステートメントの使用例](#)

# 31 構文の概要

---

■ 全般的な構文 .....	194
■ 基本的な構文要素 .....	194

この章には、DEFINE DATAステートメントの説明で使用される構文ボックスの完全な概要が記載されています。

プログラムのステートメント行の中でキーワード、節、パラメータ、オプション、および他の構文要素を配列および組み合わせる方法について説明します。

## 全般的な構文

---

```
DEFINE DATA
  [GLOBAL USING global-data-area [WITH block[.block] ...]]
  [
    PARAMETER節 { USING parameter-data-area } ] ...
                  { parameter-data-definition... } ]
  [
    OBJECT { USING { local-data-area } } ] ...
              { parameter-data-area } } ]
              { data-definition... } ]
  [
    LOCAL { USING { local-data-area } } ] ...
              { parameter-data-area } } ]
              { data-definition... } ]
  [INDEPENDENT AIV-data-definition ...]
  [
    CONTEXT { USING { local-data-area } } ]
               { parameter-data-area } } ]
               { context-data-definition ... } ]
END-DEFINE
```

## 基本的な構文要素

---

以下では次のトピックについて説明します。

- `data-definition`
- `parameter-data-definition`
- `parameter-handle-definition`
- `variable-definition`
- `view-definition`
- `redefinition`
- `init-definition`
- `array-definition`
- `array-init-definition`
- `emhdpm`

- AIV-data-definition
- context-data-definition

### data-definition

```

{
  level {
    group-name [(array-definition)]
    variable-definition
    view-definition
    redefinition
    handle-definition
  }
}

```

詳細については、「ローカルデータの定義」または「[NaturalIX オブジェクトの定義](#)」を参照してください。

### parameter-data-definition

```

{
  level {
    group-name [(array-definition)] redefinition
    variable-name {
      (format-length [/ array-definition])
      ( { A } [/array-definition] ) DYNAMIC } [BY VALUE [RESULT]] [OPTIONAL]
    parameter-handle-definition [BY VALUE [RESULT]] [OPTIONAL]
  }
}

```

詳細については、「[パラメータデータの定義](#)」を参照してください。

### parameter-handle-definition

```

handle-name [(array-definition)] HANDLE OF OBJECT

```

詳細については、[パラメータハンドルのデータ定義](#)に関する説明を参照してください。

### variable-definition

```

{
  <scalar-definition>
  <array-definition>
}

```

## 構文の概要

<scalar-definition>

$$variable-name \left\{ \begin{array}{l} (format-length) \\ \left( \left( \left( \begin{array}{c} A \\ U \\ B \end{array} \right) \right) \right) DYNAMIC \end{array} \right\} \left[ \left[ \begin{array}{c} CONSTANT \\ INIT \end{array} \right] init-definition \right] [emhdpm]$$

<array-definition>

$$variable-name \left\{ \begin{array}{l} (format-length / array-definition) \\ \left( \left( \left( \begin{array}{c} A \\ U \\ B \end{array} \right) /array-definition \right) \right) DYNAMIC \end{array} \right\} \left[ \begin{array}{c} CONSTANT \\ INIT \end{array} \right] array-init-definition ] [emhdpm]$$

詳細については、「[変数定義](#)」を参照してください。

## view-definition

$$\begin{array}{l} view-name \\ VIEW [OF] \\ ddm-name \end{array} \left[ \begin{array}{l} level \left\{ \begin{array}{l} ddm-field \left[ \begin{array}{l} ([format-length] \\ [/array-definition]) \end{array} \right] [emhdpm] \\ \left( \left( \left( \begin{array}{c} A \\ B \end{array} \right) \right) \right) DYNAMIC \end{array} \right\} \end{array} \right] \dots$$

詳細については、「[ビューの定義](#)」を参照してください。

## redefinition

$$REDEFINE field-name \left\{ \begin{array}{l} rgroup \\ level \left\{ \begin{array}{l} rfield (format-length [/array-definition]) \\ FILLER nX \end{array} \right\} \end{array} \right\} \dots$$

詳細については、「[再定義](#)」を参照してください。

**init-definition**

```

{
  <constant>
  <system-variable>
  FULL LENGTH <character-s>
  LENGTH n <character-s>
}

```

詳細については、「[配列用の初期値／定数値](#)」を参照してください。

**array-definition**

```
{bound[:bound]},...3
```

詳細については、「[配列の次元の定義](#)」を参照してください。

**array-init-definition**

```

{
  [
    [
      ALL
      index[:index]
      ( { v } ,...3 )
    ]
  ]
  {
    FULL LENGTH
    LENGTH n
    constant
    system-variable...
  } <character-s,...>
} ...

```

詳細については、「[配列用の初期値／定数値](#)」を参照してください。

**emhdpm**

```
[(EM=value) [HD='text'] [PM=value)]
```

詳細については、「[フィールド／変数のEM、HD、PMパラメータ](#)」を参照してください。

**AIV-data-definition**

```

level {
  variable-definition
  redefinition
  handle-definition
}

```

詳細については、「[アプリケーションに依存しない変数の定義](#)」を参照してください。

### context-data-definition

<i>level</i> { <i>variable-definition</i> <i>redefinition</i> <i>handle-definition</i> }
--

詳細については、「[Natural RPC 用のコンテキスト変数の定義](#)」を参照してください。



# 32

## DEFINE DATA - 全般

---

▪ 機能 .....	200
▪ 規則 .....	200
▪ プログラミングモード .....	200
▪ 詳細な情報 .....	201

## 機能

---

DEFINE DATA ステートメントは、ローカルデータエリア (LDA)、グローバルデータエリア (GDA) またはパラメータデータエリア (PDA) に含まれる定義済みのデータ定義を参照するか、インライン定義を書き込むことにより、Natural プログラムで使用するデータ定義を宣言する多数の節を提供します。

## 規則

---

- DEFINE DATA ステートメントを使用する場合は、プログラムまたはルーチンの最初のステートメントとして指定する必要があります。
- 「空の」 DEFINE DATA ステートメントを使用することはできません。つまり、少なくとも 1 つの節 (LOCAL、GLOBAL、PARAMETER、INDEPENDENT、CONTEXT、または OBJECT) を指定する必要があり、また少なくとも 1 つのフィールドを定義する必要があります。
- 複数の節を指定できますが、その場合は、構文図に示されている順序で節を指定する必要があります。
- Natural 予約語 END-DEFINE を使用して、DEFINE DATA ステートメントを終了させる必要があります。

## プログラミングモード

---

DEFINE DATA ステートメントは、ストラクチャードモードおよびレポーティングモードで使用できます。これらのモードでの違いについては、DEFINE DATA ステートメントの説明を参照してください。

通常、次のことが適用されます。

- **ストラクチャードモード**

## ■ レポートモード

### ストラクチャードモード

使用するすべての変数は、**アプリケーションに依存しない変数**（AIV）を除き、DEFINE DATA ステートメントで定義する必要があり、プログラム内の他の場所で定義することはできません。**DEFINE DATA INDEPENDENT** ステートメントを使用する場合は、プログラム内の他の場所で AIV を定義することはできません。

### レポートモード

変数はプログラムの本体で定義できるため、DEFINE DATA ステートメントは必須ではありません。ただし、**DEFINE DATA LOCAL** ステートメントをレポートモードで使用する場合は、アプリケーションに依存しない変数（AVI）を除く変数を、プログラム内の他の場所で定義することはできません。また、**DEFINE DATA INDEPENDENT** ステートメントを使用する場合は、**アプリケーションに依存しない変数**（AIV）をプログラム内の他の場所で定義することはできません。

## 詳細な情報

DEFINE DATA ステートメントの詳細については、『プログラミングガイド』の次のセクションを参照してください。

- DEFINE DATA ステートメントの使用と構造
- データエリアの使用



# 33 ローカルデータの定義

---

■ 機能 .....	204
■ 制限事項 .....	204
■ 構文説明 .....	204

## ローカルデータの定義

DEFINE DATA LOCAL の一般的な構文：

```
[ LOCAL { USING { local-data-area } } ] ...  
[ LOCAL { USING { parameter-data-area } } ] ...  
[ LOCAL { data-definition ... } ] ...
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

DEFINE DATA LOCAL ステートメントでは、アプリケーション内にある単一の Natural モジュールによって排他的に使用されるデータ要素を定義します。これらのデータ要素またはフィールドは、ステートメント自体の中で定義できます（[data-definition](#)を参照）。または、データエリアを参照するステートメントを使用して、プログラムの外にある別のローカルデータエリア（LDA）やパラメータデータエリア（PDA）でデータ要素またはフィールドを定義することもできます。

## 制限事項

LDA とそれを参照するオブジェクトは、同じライブラリ（または STEPLIB）に含まれている必要があります。

## 構文説明

<i>local-data-area</i>	<p>ローカルデータエリアには、DEFINE DATA LOCAL ステートメントに含めることができる定義済みのデータ要素が格納されます。複数のデータエリアを参照できますが、その場合は次の例のように、予約語 LOCAL および USING を繰り返す必要があります。</p> <pre>DEFINE DATA LOCAL   LOCAL USING DATX_L   LOCAL USING DATX_P    ... END-DEFINE ;</pre> <p>詳細については、『プログラミングガイド』の「別のデータエリアでのフィールドの定義」、「ローカルデータエリア」、および「例2」を参照してください。</p>
------------------------	--

<i>parameter-data-area</i>	また、DEFINE DATA LOCAL ステートメントで参照するデータエリアは、パラメータデータエリア (PDA) でもかまいません。PDA を LDA として使用することにより、PDA と同じ構造を持つ LDA を作成する手間を省くことができます。
<i>data-definition</i>	下記の「 <a href="#">ダイレクトデータ定義</a> 」を参照してください。
<b>END-DEFINE</b>	Natural 予約語 END-DEFINE を使用して、DEFINE DATA ステートメントを終了させる必要があります。

## ダイレクトデータ定義

ローカルデータは、プログラムまたはルーチン内で直接定義できます。ダイレクトデータ定義には、次の構文が適用されます。

{ level {	<i>group-name</i> [( <i>array-definition</i> )]	} }
	<i>variable-definition</i>	
	<i>view-definition</i>	
	<i>redefinition</i>	
	<i>handle-definition</i>	

詳細については、次を参照してください。

- **例1 - DEFINE DATA LOCAL** (ダイレクトデータ定義)
- 『プログラミングガイド』の「[DEFINE DATA ステートメント内のフィールドの定義](#)」
- 『プログラミングガイド』の「[ローカルデータエリア](#)」と「[例1](#)」

### ダイレクトデータ定義の構文要素の説明

<i>level</i>	<p>レベル番号は 01～99 の範囲内の 1 桁または 2 桁の数字 (先頭の 0 は任意) を使用し、フィールドグルーピングと合わせて使用します。02 またはそれ以上のレベル番号が割り当てられたフィールドは、それより小さいレベル番号が割り当てられた直前のグループの一部であるとみなされます。</p> <p>グループを定義すると、グループ名を使用して一連のフィールド (または 1 つのフィールドのみ) を参照できます。特定のステートメント (CALL、CALLNAT、RESET、WRITE など) の場合、グループ名をショートカットとして指定し、グループに含まれるフィールドを参照することもできます。</p> <p>グループには、他のグループを含めることができます。グループにレベル番号をつけるときは、レベル番号を飛ばしてはなりません。</p> <p>ビューの定義は常にレベル 1 で定義する必要があります。</p>
<i>group-name</i>	<p>グループの名前。名前は、Natural 変数名の定義ルールに準拠していなければなりません。次のセクションも参照してください。</p>

## ローカルデータの定義

	<ul style="list-style-type: none"><li>■ 『Natural の使用』ドキュメントの「ユーザー定義変数の命名規則」</li><li>■ 『プログラミングガイド』の「データ構造の条件指定」</li></ul>
<i>array-definition</i>	<i>array-definition</i> の場合、配列定義の次元の下限と上限を定義します。「 <a href="#">配列の次元の定義</a> 」を参照してください。
<i>variable-definition</i>	<i>variable-definition</i> を使用して、単一の値（スカラ）または複数の値（配列）を持つ単一のフィールド／変数を定義します。「 <a href="#">変数定義</a> 」を参照してください。
<i>view-definition</i>	<i>view-definition</i> を使用して、データ定義モジュール（DDM）から派生したビューを定義します。「 <a href="#">ビューの定義</a> 」を参照してください。
<i>redefinition</i>	<i>redefinition</i> を使用して、グループ、ビュー、DDM フィールド、単一のフィールド／変数（スカラまたは配列）を再定義できます。「 <a href="#">再定義</a> 」を参照してください。
<i>handle-definition</i>	ハンドルは、コードのダイアログエレメントを特定し、ハンドル変数に格納されます。「 <a href="#">ハンドルの定義</a> 」を参照してください。



# 34 グローバルデータの定義

---

- 機能 ..... 208
- 構文説明 ..... 208

## グローバルデータの定義

---

DEFINE DATA GLOBAL の一般的な構文：

```
DEFINE DATA  
GLOBAL USING global-data-area [WITH block [.block...]]  
END-DEFINE
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

---

DEFINE DATA GLOBAL ステートメントでは、[グローバルデータエリア](#)（GDA）を使用してデータ要素を定義します。

## 構文説明

---

<b>USING</b> <i>global-data-area</i>	グローバルデータエリア（GDA）には、複数のプログラミングオブジェクトで参照できるデータ要素が含まれています。
<b>WITH</b> <i>block</i>	データストレージスペースを保存するために、データブロックでグローバルデータエリアを作成できます。プログラムの実行中にデータブロックが互いに重なり合うことも可能であるため、ストレージスペースの節約になります。  ブロックレベルの最大値は 8（マスタブロックを含む）です。詳細については、『 <a href="#">プログラミングガイド</a> 』の「 <a href="#">データブロック</a> 」を参照してください。
<i>.block</i>	<i>.block</i> 表記（複数可）では、プログラム内で使用されるブロックを指定できます。
<b>END-DEFINE</b>	Natural 予約語 END-DEFINE を使用して、DEFINE DATA ステートメントを終了させる必要があります。

# 35      パラメータデータの定義

---

▪ 機能 .....	210
▪ 制限事項 .....	210
▪ 構文説明 .....	210

## パラメータデータの定義

DEFINE DATA PARAMETER の一般的な構文：

```
[ PARAMETER { USING parameter-data-area  
parameter-data-definition... } ] ...
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

DEFINE DATA PARAMETER ステートメントでは、Natural サブプログラム、外部サブルーチン、またはヘルプルーチンで、受信パラメータとして使用できるデータ要素を定義します。これらのパラメータは、ステートメント自体の中で定義できます（下記の「[パラメータデータ定義](#)」を参照）。または、データエリアを参照するステートメントを使用して、プログラムの外にあるパラメータデータエリア（PDA）でパラメータを定義することもできます。

## 制限事項


- パラメータのデータ要素には、初期値または定数値を割り当てることはできず、またこれらのデータ要素では、編集マスク（EM）、ヘッダー（HD）、または出力モード（PM）を保持することはできません（「[フィールド／変数の EM、HD、PM パラメータ](#)」も参照）。
- パラメータデータエリアとそれを参照するオブジェクトは、同じライブラリ（または STEPLIB）に含まれている必要があります。

## 構文説明

<b>USING</b> <i>parameter-data-area</i>	サブプログラム、外部サブルーチン、またはダイアログでパラメータとして使用されるデータ要素を含む <i>parameter-data-area</i> の名前。
<i>parameter-data-definition</i>	パラメータデータエリアを定義する代わりに、プログラムまたはサブルーチン内でパラメータデータを直接定義することもできます。下記の「 <a href="#">パラメータデータ定義</a> 」を参照してください。
<b>END-DEFINE</b>	Natural 予約語 END-DEFINE を使用して、DEFINE DATA ステートメントを終了させる必要があります。

## パラメータデータ定義

ダイレクトパラメータデータ定義には、次の構文が適用されます。



構文要素の説明：

<i>level</i>	<p>レベル番号は 01~99 の範囲内の 1 桁または 2 桁の数字（先頭の 0 は任意）を使用し、フィールドグルーピングと合わせて使用します。02 またはそれ以上のレベル番号が割り当てられたフィールドは、それより小さいレベル番号が割り当てられた直前のグループの一部であるとみなされます。</p> <p>グループを定義すると、グループ名を使用して一連のフィールド（または 1 つのフィールドのみ）を参照できます。特定のステートメント（CALL、CALLNAT、RESET、WRITE など）の場合、グループ名をショートカットとして指定し、グループに含まれるフィールドを参照することもできます。</p> <p>グループには、他のグループを含めることができます。グループにレベル番号をつけるときは、レベル番号を飛ばしてはなりません。</p>
<i>group-name</i>	<p>グループの名前。名前は、Natural 変数名の定義ルールに準拠していなければなりません。次のセクションも参照してください。</p> <ul style="list-style-type: none"> <li>■ 『Natural の使用』ドキュメントの「ユーザー定義変数の命名規則」</li> <li>■ 『プログラミングガイド』の「データ構造の条件指定」</li> </ul>
<i>array-definition</i>	<p><i>array-definition</i> の場合、配列定義の次元の下限と上限を定義します。「<a href="#">配列の次元の定義</a>」および「<a href="#">パラメータデータエリア内の可変配列</a>」を参照してください。</p>
<i>redefinition</i>	<p><i>redefinition</i> を使用して、グループまたは単一のフィールド／変数（スカラーまたは配列）を再定義できます。「<a href="#">再定義</a>」を参照してください。</p> <p><b>注意:</b> <i>parameter-data-definition</i> では、グループの「再定義」は REDEFINE ブロック内のみで行うことができます。</p>

<i>variable-name</i>	変数に割り当てられる名前。Natural 変数名のルールが適用されま す。ユーザー定義変数の命名規則については、『Natural の使用』ド キュメントの「ユーザー定義変数の命名規則」を参照してください。
<i>format-length</i>	フィールドのフォーマットおよび長さ。ユーザー定義変数のフォー マットおよび長さの定義については、『プログラミングガイド』の 「ユーザー定義変数のフォーマットおよび長さ」を参照してくださ い。
<b>A、U、または B</b>	データタイプ：ダイナミック変数用の英数字（A）、Unicode（U）、 またはバイナリ（B）。
<b>DYNAMIC</b>	パラメータは DYNAMIC として定義できます。ダイナミック変数の処 理の詳細については、「ダイナミック変数およびフィールドについ て」を参照してください。
	コールモード：  参照による呼び出し、値による呼び出し、値による呼び出しと結果の いずれを使用するかに応じて、適切な転送メカニズムが適用されま す。詳細については、CALLNAT ステートメントを参照して下さい。
<b>(BY VALUE を省略)</b>	参照による呼び出し：  参照による呼び出しは、BY VALUE キーワードを省略した場合にデ フォルトで有効になります。この場合、パラメータは参照によって （つまりアドレスを介して）サブプログラムまたはサブルーチンに渡 されるため、CALLNAT または PERFORM ステートメント内でパラメー タとして指定するフィールドのフォーマット/長さは、サブプログラ ムまたはサブルーチンで呼び出される対応するフィールドと同じであ る必要があります。
<b>BY VALUE</b>	値による呼び出し：  BY VALUE を指定すると、パラメータは値によってサブプログラムま たはサブルーチンに渡されます。つまり、（アドレスの代わりに）実 際のパラメータ値が渡されます。さらに、サブプログラムまたはサ ブルーチン内にあるフィールドのフォーマット/長さは、CALLNAT または PERFORM パラメータと同じである必要はありません。フォー マット/長さでは、データ転送の互換性のみが必要です。データ転 送の互換性に関しては、『プログラミングガイド』の「演算割り当て のルール」セクションにある「データ転送」の対応表が適用されま す。  BY VALUE を使用すると、例えば、サブプログラムまたはサブルーチ ンを呼び出すオブジェクトを調整しなくても、サブプログラムまたは サブルーチン内でフィールドの長さを増加させることができます（サ ブプログラムまたはサブルーチンの拡張によってそれが必要になった 場合）。  BY VALUE の例：

	<pre>* Program DEFINE DATA LOCAL 1 #FIELDA (P5) ... END-DEFINE ... CALLNAT 'SUBR01' #FIELDA ...</pre>	<pre>* Subroutine SUBR01 DEFINE DATA PARAMETER 1 #FIELDB (P9) BY VALUE END-DEFINE ...</pre>
<b>BY VALUE RESULT</b>	<p>値による呼び出しと結果：</p> <p>BY VALUEはサブプログラムまたはサブルーチンに渡されるパラメータに適用されますが、BY VALUE RESULTでは、パラメータが値によって両方向に渡されます。つまり、実際のパラメータ値が呼び出し側のオブジェクトからサブプログラムまたはサブルーチンに渡され、呼び出し側のオブジェクトに戻るときに、実際のパラメータ値がサブプログラムまたはサブルーチンから呼び出し側オブジェクトに返されます。</p> <p>BY VALUE RESULTを指定する場合は、対象フィールドのフォーマット／長さに関して、両方向のデータ転送で互換性を確保する必要があります。</p>	
<b>OPTIONAL</b>	<p>OPTIONALを使用しないで定義したパラメータ（デフォルト）には、呼び出し側のオブジェクトから値を渡す必要があります。</p> <p>OPTIONALを使用して定義したパラメータには、値を渡すことはできますが、呼び出し側のオブジェクトからこのパラメータに値を渡す必要はありません。</p> <p>呼び出し側のオブジェクトでは、スキップされるパラメータが、表記 <i>nX</i> を使用して示されます。つまり、値は渡されません。</p> <p>SPECIFIED オプションを指定すると、実行時にオプションパラメータが定義されているかどうかを確認できます。</p>	
<i>parameter-handle-definition</i>	<p>下記の「<a href="#">パラメータハンドルの定義</a>」を参照してください。</p>	

### パラメータハンドルの定義

*parameter-handle-definition* の構文：

```
handle-name [(array-definition)] HANDLE OF OBJECT
```

構文要素の説明：

<i>handle-name</i>	ハンドルに割り当てる名前です。ユーザー定義変数の命名規則が適用されます。『 <i>Natural</i> の使用』ドキュメントの「ユーザー定義変数の命名規則」を参照してください。
<b>HANDLE OF OBJECT</b>	『プログラミングガイド』の「 <i>NaturalX</i> 」セクションで説明されているように、 <i>NaturalX</i> と一緒に使用します。
<i>array-definition</i>	<i>array-definition</i> の場合、配列定義の次元の下限と上限を定義します。「 <a href="#">配列の次元の定義</a> 」を参照してください。



## 36 アプリケーションに依存しない変数の定義

---

- 機能 ..... 216
- 構文説明 ..... 216

## アプリケーションに依存しない変数の定義

---

DEFINE DATA INDEPENDENT の一般的な構文：

```
DEFINE DATA  
INDEPENDENT AIV-data-definition...  
END-DEFINE
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。


## 機能

---

DEFINE DATA INDEPENDENT ステートメントでは、アプリケーションに依存しない変数 (AIV) を定義します。

アプリケーションに依存しない変数は、その名前によって参照されます。またはこの変数の内容は、その名前を参照する 1 つのアプリケーション内で実行されるすべてのプログラミングオブジェクトによって共有されます。変数は、この変数を参照する最初のプログラミングオブジェクトによって割り当てられ、LOGON コマンドまたは [RELEASE VARIABLES](#) ステートメントによって解除されます。

任意の [INIT](#) 節は、その節を含む、各実行プログラミングオブジェクトで評価されます (変数を割り当てるプログラミングオブジェクト内だけではありません)。

 **注意:** RPC サーバーでは、アプリケーションに依存しない変数 (AIV) は、暗黙的に割り当て解除されず、複数の RPC 要求にわたってアクティブになります。これは、RPC サーバー上の同じ変数に複数のクライアントがアクセスする場合があります。このため、割り当てを解除するには、明示的に [RELEASE VARIABLES](#) ステートメントを使用する必要があります。『[Natural リモートプロシージャコール](#)』ドキュメントの「[アプリケーションに依存しない変数](#)」も参照してください。

## 構文説明

---

<b>INDEPENDENT</b> <i>AIV-data-definition</i>	DEFINE DATA INDEPENDENT ステートメントでは、1 つまたは複数のアプリケーションに依存しない変数 (AIV) を定義できます。各 AIV には、 <a href="#">下記</a> の構文が適用されます。
<b>END-DEFINE</b>	Natural 予約語 END-DEFINE を使用して、DEFINE DATA ステートメントを終了させる必要があります。

## AIV データ定義

<i>level</i>	$\left\{ \begin{array}{l} \textit{variable-definition} \\ \textit{redefinition} \\ \textit{handle-definition} \end{array} \right\}$
--------------	---

構文要素の説明：

<i>level</i>	アプリケーションに依存しない変数は、レベル01で定義する必要があります。他のレベルは再定義だけで使用します。
<i>variable-definition</i>	<p><i>variable definition</i>を使用して、単一の値（スカラー）または複数の値（配列）を持つ単一のフィールド／変数を定義します。「<a href="#">変数定義</a>」を参照してください。</p> <p><b>注意:</b> アプリケーションに依存しない変数の名前は、プラス (+) 文字で始める必要があります。</p>
<i>redefinition</i>	<p><i>redefinition</i>を使用して、グループ、ビュー、DDM フィールド、単一のフィールド／変数（スカラーまたは配列）を再定義できます。「<a href="#">再定義</a>」を参照してください。</p> <p>再定義によって生成されるフィールドは、アプリケーションに依存しない変数にすることはできません。つまり、これらのフィールドの名前を、プラス (+) で始めることはできません。これらのフィールドは、ローカル変数として処理されます。</p>
<i>handle-definition</i>	ハンドルは、コードのダイアログエレメントを特定し、ハンドル変数に格納されます。「 <a href="#">ハンドルの定義</a> 」を参照してください。



**注意:** 名前の最初の文字は、プラス (+) にする必要があります。Natural 変数の名前に適用される規則については、『[Natural の使用](#)』ドキュメントの「[ユーザー定義変数の命名規則](#)」を参照してください。




# 37 Natural RPC 用のコンテキスト変数の定義

---

▪ 機能 .....	220
▪ 制限事項 .....	221
▪ 構文説明 .....	221

DEFINE DATA CONTEXT の一般的な構文：



構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[Natural リモートプロシージャコール](#)」

## 機能

DEFINE DATA CONTEXT ステートメントは、Natural リモートプロシージャコール (RPC) と一緒に使用します。このステートメントは、コンテキスト変数と呼ばれる変数の定義に使用します。コンテキスト変数では、対応する CALLNAT ステートメントで変数をパラメータとして明示的に渡さなくても、1つの会話の中で複数のリモートサブプログラムがこの変数を使用できます。

コンテキスト変数は、その名前によって参照されます。また、この変数の内容は、その名前を参照する1つの会話で実行されるすべてのプログラミングオブジェクトによって共有されます。変数は、変数の定義を含むプログラミングオブジェクトが最初に実行されたときに割り当てられ、会話が終了すると解除されます。

コンテキスト変数は、非会話型 CALLNAT で使用されます。この場合、コンテキスト変数は非会話型 CALLNAT が単独で呼び出されている間だけ存在していますが、CALLNAT から呼び出される側のすべてで共有できます。

コンテキスト変数は、会話内で呼び出されたサブプログラムで共有されることはありません。このようなサブプログラムまたはサブプログラムから呼び出される側の1つでコンテキスト変数を参照する場合は、その変数用に別のストレージエリアが割り当てられます。

任意の INIT 節は、その節を含む、各実行プログラミングオブジェクトで評価されます（変数を割り当てるプログラミングオブジェクト内だけではありません）。これは、グローバル変数の INIT の動作とは異なります。

詳細については、『[Natural リモートプロシージャコール \(RPC\)](#)』ドキュメントの「[会話コンテキストの定義](#)」を参照してください。

## 制限事項

コンテキスト変数は、レベル 01 で定義する必要があります。他のレベルは再定義だけで使用します。

## 構文説明

<b>USING</b> <i>local-data-area</i>	<p>ローカルデータエリア（LDA）には、単一のNaturalモジュールで使用されるデータ要素が含まれています。複数のデータエリアを参照できますが、その場合は、予約語 CONTEXT と USING を繰り返す必要があります。例えば、次のようになります。</p> <pre>DEFINE DATA   CONTEXT USING DATX_L   CONTEXT USING DATX_P   ... END-DEFINE ;</pre> <p>詳細については、『プログラミングガイド』の「別のデータエリアでのフィールドの定義」を参照してください。</p>
<b>USING</b> <i>parameter-data-area</i>	<p>パラメータデータエリアには、サブプログラム、外部サブルーチン、またはダイアログでパラメータとして使用されるデータ要素が含まれています。</p>
<i>context-data-definition</i>	<p>コンテキストデータは、プログラムまたはルーチン内で直接定義することができます。ダイレクトデータ定義には、<a href="#">下記</a>の構文が適用されます。</p>
<b>END-DEFINE</b>	<p>Natural 予約語 END-DEFINE を使用して、DEFINE DATA ステートメントを終了させる必要があります。</p>

### コンテキストデータ定義

コンテキストデータは、プログラムまたはルーチン内で直接定義することができます。ダイレクトデータ定義には、次の構文が適用されます。

<i>level</i>	$\left\{ \begin{array}{l} \textit{variable-definition} \\ \textit{redefinition} \\ \textit{handle-definition} \end{array} \right\}$
--------------	---

詳細については、『プログラミングガイド』の「*DEFINE DATA* ステートメント内のフィールドの定義」を参照してください。

<i>level</i>	アプリケーションに依存しない変数は、レベル01で定義する必要があります。他のレベルは再定義だけで使用します。
<i>variable-definition</i>	<i>variable-definition</i> を使用して、単一の値（スカラー）または複数の値（配列）を持つ単一のフィールド／変数を定義します。「 <a href="#">変数定義</a> 」を参照してください。  <b>注意:</b> <i>CONSTANT</i> 節は、このコンテキストでは使用できません。
<i>redefinition</i>	<i>redefinition</i> を使用して、グループ、ビュー、DDM フィールド、単一のフィールド／変数（スカラーまたは配列）を再定義できます。「 <a href="#">再定義</a> 」を参照してください。
<i>handle-definition</i>	ハンドルは、コードのダイアログエレメントを特定し、ハンドル変数に格納されます。「 <a href="#">ハンドルの定義</a> 」を参照してください。



**注意:** 再定義によって生成されるフィールドは、コンテキスト変数とはみなされません。これらのフィールドは、ローカル変数として処理されます。




# 38 NaturalX オブジェクトの定義

---

- 機能 ..... 224
- 構文説明 ..... 224

DEFINE DATA OBJECT の一般的な構文：



構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能


DEFINE DATA OBJECT ステートメントは、サブプログラムまたはクラス内で、NaturalX と一緒に使用します。詳細については、『[プログラミングガイド](#)』の「[NaturalX](#)」セクションを参照してください。

## 構文説明

<b>USING</b>  <i>local-data-area</i>	<p>ローカルデータエリア (LDA) には、単一の Natural モジュールで使用されるデータ要素が含まれています。複数のデータエリアを参照できますが、その場合は次の例のように、予約語 OBJECT および USING を繰り返す必要があります。</p> <pre>DEFINE DATA   OBJECT USING DATX_L   OBJECT USING DATX_P    ... END-DEFINE ;</pre> <p>詳細については、『<a href="#">プログラミングガイド</a>』の「<a href="#">別のデータエリアでのフィールドの定義</a>」を参照してください。</p>
<b>USING</b>  <i>parameter-data-area</i>	<p>DEFINE DATA OBJECT ステートメントで定義するデータエリアは、パラメータデータエリア (PDA) でもかまいません。PDA をオブジェクトデータエリアとして使用することにより、PDA と同じ構造を持つオブジェクトデータエリアを作成する手間を省くことができます。</p>
<i>data-definition</i>	<p>また、下記の「<a href="#">ダイレクトデータ定義</a>」セクションで示されている構文を使用して、データを直接定義することもできます。</p>
<b>END-DEFINE</b>	<p>Natural 予約語 END-DEFINE を使用して、DEFINE DATA ステートメントを終了させる必要があります。</p>

## ダイレクトデータ定義

また、次の構文を使用してデータを直接定義することもできます。



詳細については、『プログラミングガイド』の「*DEFINE DATA* ステートメント内のフィールドの定義」を参照してください。

<i>level</i>	<p>レベル番号は 01～99 の範囲内の 1 桁または 2 桁の数字（先頭の 0 は任意）を使用し、フィールドグルーピングと合わせて使用します。02 またはそれ以上のレベル番号が割り当てられたフィールドは、それより小さいレベル番号が割り当てられた直前のグループの一部であるとみなされます。</p> <p>グループを定義すると、グループ名を使用して一連のフィールド（または 1 つのフィールドのみ）を参照できます。特定のステートメント（CALL、CALLNAT、RESET、WRITE など）の場合、グループ名をショートカットとして指定し、グループに含まれるフィールドを参照することもできます。</p> <p>グループには、他のグループを含めることができます。グループにレベル番号をつけるときは、レベル番号を飛ばしてはなりません。</p> <p>ビューの定義は常にレベル 1 で定義する必要があります。</p>
<i>group-name</i>	<p>グループの名前。名前は、Natural 変数名の定義ルールに準拠していなければなりません。次のセクションも参照してください。</p> <ul style="list-style-type: none"> <li>■ 『Natural の使用』ドキュメントの「ユーザー定義変数の命名規則」</li> <li>■ 『プログラミングガイド』の「データ構造の条件指定」</li> </ul>
<i>array-definition</i>	<p><i>array-definition</i> の場合、配列定義の次元の下限と上限を定義します。「<a href="#">配列の次元の定義</a>」を参照してください。</p>
<i>variable-definition</i>	<p><i>variable-definition</i> を使用して、単一の値（スカラー）または複数の値（配列）を持つ単一のフィールド／変数を定義します。「<a href="#">変数定義</a>」を参照してください。</p>
<i>view-definition</i>	<p><i>view-definition</i> を使用して、データ定義モジュール（DDM）から派生したビューを定義します。「<a href="#">ビューの定義</a>」を参照してください。</p>
<i>redefinition</i>	<p><i>redefinition</i> を使用して、グループ、ビュー、DDM フィールド、単一のフィールド／変数（スカラーまたは配列）を再定義できます。「<a href="#">再定義</a>」を参照してください。</p>
<i>handle-definition</i>	<p>ハンドルは、コードのダイアログエレメントを特定し、ハンドル変数に格納されます。「<a href="#">ハンドルの定義</a>」を参照してください。</p>



# 39 変数定義

---

- 機能 ..... 228
- 構文説明 ..... 228

## 変数定義

DEFINE DATA LOCAL、DEFINE DATA INDEPENDENT、DEFINE DATA CONTEXT、および DEFINE DATA OBJECT で使用する *variable-definition* では、*scalar-definition* または *array-definition* のいずれかを指定できます。

```
{ <scalar-definition> }  
{ <array-definition> }
```

<scalar-definition>

```
variable-name { (format-length) } [ { { { A } } } DYNAMIC ] [ { { CONSTANT } } init-definition ] [emhdpm]  
              { ( { { U } } ) }  
              { { { B } } }
```

<array-definition>

```
variable-name { (format-length/ array-definition) } [ { { { A } } } DYNAMIC ] [ { { CONSTANT } } array-init-definition ] [emhdpm]  
              { ( { { U } } /array-definition) }  
              { { { B } } }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

*variable-definition* を使用して、単一の値（スカラー）または複数の値（配列）を持つ単一のフィールド／変数を定義します。

## 構文説明

<i>variable-name</i>	変数に割り当てられる名前。Natural変数名のルールが適用されます。DEFINE DATA INDEPENDENT で使用する場合、変数名はプラス文字 (+) で始まる必要があります。  ユーザー定義変数の命名規則については、『 <a href="#">Naturalの使用</a> 』ドキュメントの「 <a href="#">ユーザー定義変数の命名規則</a> 」を参照してください。
<i>format-length</i>	フィールドのフォーマットおよび長さ。ユーザー定義変数のフォーマットおよび長さの定義については、『 <a href="#">プログラミングガイド</a> 』の「 <a href="#">ユーザー定義変数のフォーマットおよび長さ</a> 」を参照してください。

<b>A、U、またはB</b>	データタイプ：ダイナミック変数用の英数字（A）、Unicode（U）、またはバイナリ（B）。
<i>array-definition</i>	<i>array-definition</i> の場合、配列定義の次元の下限と上限を定義します。「 <a href="#">配列の次元の定義</a> 」を参照してください。
<b>DYNAMIC</b>	フィールドは、DYNAMICとして定義できます。ダイナミック変数の処理については、「 <a href="#">ラージ変数／フィールドとダイナミック変数／フィールド</a> 」を参照してください。
<b>CONSTANT</b>	<p>変数／配列は、指定された定数として処理されます。割り当てられた定数値は、変数／配列が参照されるたびに使用されます。割り当てられた値は、プログラムの実行中には変更できません。</p> <p>『プログラミングガイド』の「<a href="#">フィールドの定義</a>」、「<a href="#">ユーザー定義定数</a>」、および「<a href="#">名前付き定数の定義</a>」も参照してください。</p> <p><b>注意:</b> 内部処理の理由から、1つのグループ定義内で変数定義と定数定義を混在させることは許可されていません。つまり、1つのグループに含めることができるのは、変数か定数のいずれかのみです。CONSTANT節を <b>DEFINE DATA INDEPENDENT</b> および <b>DEFINE DATA CONTEXT</b> で使用することはできません。CONST節を X-array と一緒に使用することはできません。</p>
<b>INIT</b>	<p>変数／配列に初期値が割り当てられます。この値は、変数／配列が <b>RESET INITIAL</b> ステートメントで参照されるときにも使用されます。</p> <p>INIT 指定をしない場合、フィールドはそのフォーマットに対応するデフォルトの初期値で初期化されます（下記の「<a href="#">デフォルトの初期値</a>」を参照）。</p> <p>『プログラミングガイド』の「<a href="#">フィールドの定義</a>」の「<a href="#">初期値</a>」も参照してください。</p> <p><b>注意:</b> <b>DEFINE DATA INDEPENDENT</b> および <b>DEFINE DATA CONTEXT</b> では、INIT節は、その節を含む、各実行プログラミングオブジェクトで評価されます（変数を割当てるプログラミングオブジェクト内だけではありません）。これは、グローバル変数のINITの動作とは異なります。INIT節を X-array と一緒に使用することはできません。</p>
<i>init-definition</i>	<i>init-definition</i> オプションを使用して、変数の初期値および定数値を定義します。「 <a href="#">初期値の定義</a> 」を参照してください。
<i>array-init-definition</i>	<i>array-init-definition</i> を使用して、配列の初期値および定数値を定義します。「 <a href="#">配列用の初期値／定数値</a> 」を参照してください。
<i>emhdpm</i>	このオプションを使用して、フィールド／変数に有効な追加のパラメータを定義できます。「 <a href="#">フィールド／変数のEM、HD、PMパラメータ</a> 」を参照してください。

### デフォルトの初期値

フォーマット	デフォルトの初期値
B、F、I、N、P	0
A、U	(空白)
L	FALSE
D	D' '
T	T'00:00:00'
C	(AD=D)
オブジェクトハンドル	NULL-HANDLE

DYNAMIC として宣言されたフィールドは、フィールド長がデフォルトで0なので、初期値を持ちません。




# 40 ビューの定義

---

▪ 機能 .....	232
▪ 構文説明 .....	232

## ビューの定義


DEFINE DATA LOCAL および DEFINE DATA OBJECT で使用する *view-definition* オプションの構文は、次のとおりです。



構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

*view-definition* を使用して、データ定義モジュール (DDM) から派生したデータビューを定義します。

 **注意:** パラメータデータエリアでは、*view-definition* を使用できません。

詳細については、『プログラミングガイド』の「[Adabas データベースのデータへのアクセス](#)」セクションで、特に次のトピックを参照してください。

- データ定義モジュール - DDM
- データベース配列
- DEFINE DATA ビュー

## 構文説明

<i>view-name</i>	ビューに割り当てられる名前。Natural 変数の名前に適用される規則については、『 <a href="#">Natural の使用</a> 』ドキュメントの「 <a href="#">ユーザー定義変数の命名規則</a> 」を参照してください。
VIEW [OF] <i>ddm-name</i>	ビューを取得する元の DDM の名前。
<i>level</i>	レベル番号は 01~99 の範囲内の 1 桁または 2 桁の数字（先頭の 0 は任意）を使用し、フィールドグルーピングと合わせて使用します。02 またはそれ以上のレベル番号が割り当てられたフィールドは、それより小さいレベル番号が割り当てられた直前のグループの一部であるとみなされます。

	<p>グループを定義すると、グループ名を使用して一連のフィールド（または1つのフィールドのみ）を参照できます。特定のステートメント（CALL、CALLNAT、RESET、WRITEなど）の場合、グループ名をショートカットとして指定し、グループに含まれるフィールドを参照することもできます。</p> <p>グループには、他のグループを含めることができます。グループにレベル番号をつけるときは、レベル番号を飛ばしてはなりません。</p>
<i>ddm-field</i>	<p>DDM から取得されるフィールドの名前。</p> <p><b>HISTOGRAM</b> ステートメントのビューを定義する場合、ビューには、HISTOGRAM を実行するディスクリプタのみが含まれている必要があります。</p>
<i>redefinition</i>	<p><i>redefinition</i> を使用して、グループ、ビュー、DDM フィールド、単一のフィールド/変数（スカラまたは配列）を再定義できます。「<b>再定義</b>」セクションを参照してください。</p>
<i>format-length</i>	<p>フィールドのフォーマットおよび長さ。省略した場合、これらはDDMから取得されます。</p> <p>ストラクチャードモードでは、フォーマットおよび長さの定義（提供される場合）を、DDMに指定されているフォーマットおよび長さと同じにする必要があります。</p> <p>レポートモードでは、フォーマットおよび長さの定義（提供される場合）と、DDMに指定されているフォーマットおよび長さとの間で、タイプの互換性を確保する必要があります。</p>
<b>A、U、またはB</b>	<p>データタイプ：ダイナミック変数用の英数字（A）、Unicode（U）、またはバイナリ（B）。</p> <p><b>注意:</b></p> <ol style="list-style-type: none"> <li>メインフレーム上の Adabas では、フォーマット U を LA フィールド（長さ ≤ 16341 バイト）に使用できますが、LB フィールド（長さ ≤ 1 GB）には使用できません。</li> <li>フォーマット B は、Adabas では使用できません。</li> </ol>
<i>array-definition</i>	<p>使用されているプログラミングモードによっては、配列（ピリオディックグループフィールド、マルチプルバリューフィールド）にオカレンス数に関する情報を含める必要が生じる場合があります。下記の「<b>ビューの配列定義</b>」を参照してください。</p>
<i>emhdpm</i>	<p>このオプションを使用して、フィールド/変数に有効な追加のパラメータを定義できます。「<b>フィールド/変数のEM、HD、PMパラメータ</b>」を参照してください。</p>
<b>DYNAMIC</b>	<p>ビューフィールドを DYNAMIC として定義します。ダイナミック変数の処理については、「<b>ラージ変数/フィールドとダイナミック変数/フィールド</b>」セクションを参照してください。</p>

### ビューの配列定義

使用されているプログラミングモードによっては、配列（ピリオディックグループフィールド、マルチプルバリュースフィールド）にオカレンス数に関する情報を含める必要が生じる場合があります。

- ストラクチャードモード
- レポートイングモード

#### ストラクチャードモード

配列を表すビューでフィールドを使用する場合は、次のことが適用されます。

- MU フィールドおよび PE フィールドのインデックス値を指定する必要があります。
- フォーマット／長さ指定をしない場合、値は DDM から取得されます。
- フォーマット／長さ指定をする場合は、フォーマットおよび長さを DDM のものと同じにする必要があります。

ストラクチャードモードでのデータベース固有の考慮事項：

<b>Adabas :</b>	MU フィールドおよび PE フィールド（DDM で定義）をビューの中で使用する場合は、これらのフィールドを配列の添字指定に含める必要があります。MU フィールドまたは通常の PE フィールドに対しては、例えば (1:10) のように 1 次元の添字範囲を指定します。PE グループの中にある MU フィールドに対しては、例えば (1:10,1:5) のように 2 次元の添字範囲を指定します。
-----------------	---

ストラクチャードモードの例：

```
DEFINE DATA LOCAL
1 EMP1 VIEW OF EMPLOYEES
  2 NAME(A20)
  2 ADDRESS-LINE(A20 / 1:2)

1 EMP2 VIEW OF EMPLOYEES
  2 NAME
  2 ADDRESS-LINE(1:2)

1 EMP3 VIEW OF EMPLOYEES
  2 NAME
  2 ADDRESS-LINE(2)

1 #K (I4)
1 EMP4 VIEW OF EMPLOYEES
  2 NAME
  2 ADDRESS-LINE(#K:#K+1)
END-DEFINE
END
```

## レポートイングモード

このモードでは、ストラクチャードモードと同じ規則が有効です。ただし、2つの例外があります。

- インデックス値を提供する必要はありません。この場合、不明な次元の添字範囲は (1:1) に設定されます。
- フォーマット／長さ指定は、DDMでの指定と異なる可能性があります。その場合は、フォーマットおよび長さの定義と、DDMに指定されているフォーマットおよび長さとの間で、タイプの互換性を確保する必要があります。

例：

```
DEFINE DATA LOCAL
1 EMP1 VIEW OF EMPLOYEES
  2 NAME(A30)
  2 ADDRESS-LINE(A35 / 5:10)

1 EMP2 VIEW OF EMPLOYEES
  2 NAME
  2 ADDRESS-LINE(A40)          /* ADDRESS LINE (1:1) IS ASSUMED

1 EMP3 VIEW OF EMPLOYEES
  2 NAME
  2 ADDRESS-LINE              /* ADDRESS LINE (1:1) IS ASSUMED

1 #K (I4)
1 EMP4 VIEW OF EMPLOYEES
  2 NAME
  2 ADDRESS-LINE(#K:#K+1)
END-DEFINE
END
```



# 41 再定義

---

▪ 機能 .....	238
▪ 制限事項 .....	238
▪ 構文説明 .....	239

## 再定義

---

DEFINE DATA LOCAL、DEFINE DATA PARAMETER、DEFINE DATA INDEPENDENT、DEFINE DATA CONTEXT、および DEFINE DATA OBJECT で使用する *redefinition* オプションの構文は、次のとおりです。

```
REDEFINE field-name { level { rgroup  
                             rfield(format-length [/array-definition]) } } ...  
                             FILLER nX
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

---

*redefinition* を使用して、グループ、ビュー、DDM フィールド、単一のフィールド/変数（スカラーまたは配列）を再定義できます。

### 注意:

1. ビューまたは DDM フィールドの「再定義」は、*parameter-data-definition* には適用されません。
2. Unicode フィールドを、英数字 (A) フィールドまたは数値 (N) フィールドとして再定義することはできません。

『プログラミングガイド』の「フィールドの再定義」も参照してください。

## 制限事項

---

- ハンドル、X-array、ダイナミック変数を再定義すること、および再定義節に含めることはできません。
- ハンドル、X-array、またはダイナミック変数を含むグループの再定義のみを行うことはできますが、問題の要素を含めることや超えることはできません。



## 構文説明

<i>field-name</i>	再定義するグループ、ビュー、DDM フィールド、または単一フィールドの名前。
<i>level</i>	レベル番号は 01~99 の範囲内の 1 桁または 2 桁の数字（先頭の 0 は任意）を使用し、フィールドグルーピングと合わせて使用します。02 またはそれ以上のレベル番号が割り当てられたフィールドは、それより小さいレベル番号が割り当てられた直前のグループの一部であるとみなされます。
<i>rgroup</i>	再定義によって生成されるグループの名前。 <b>注意:</b> <i>view-definition</i> 内の <i>redefinition</i> では、 <i>rgroup</i> の名前は、基礎となる DDM 内のフィールド名とは異なっている必要があります。
<i>rfield</i>	再定義によって生成されるフィールドの名前。 <b>注意:</b> <i>view-definition</i> 内の <i>redefinition</i> では、 <i>rfield</i> の名前は、基礎となる DDM 内のフィールド名とは異なっている必要があります。
<i>format-length</i>	生成されるフィールド ( <i>rfield</i> ) のフォーマットおよび長さ。
<i>array-definition</i>	<i>array-definition</i> の場合、配列定義の次元の下限と上限を定義します。「 <a href="#">配列の次元の定義</a> 」を参照してください。
<b>FILLER nX</b>	この表記では、再定義するフィールドに <i>n</i> フィラーバイト（使用しないセグメント）を定義します。末尾の充填バイトは任意指定です。

## REDEFINE の使用例

例 1:	例 2:	例 3:
<pre> DEFINE DATA LOCAL   01 #VAR1 (A15)   01 #VAR2       02 #VAR2A (N4.1) INIT &lt;0&gt;       02 #VAR2B (P6.2) INIT &lt;0&gt;   01 REDEFINE #VAR2       02 #VAR2RD (A10) END-DEFINE ... </pre>	<pre> DEFINE DATA LOCAL   01 MYVIEW VIEW OF STAFF   02 NAME   02 BIRTH   02 REDEFINE BIRTH       03 BIRTH-YEAR (N4)       03 BIRTH-MONTH (N2)       03 BIRTH-DAY (N2) END-DEFINE ... </pre>	<pre> DEFINE DATA LOCAL   1 #FIELD (A12)   1 REDEFINE #FIELD       2 #RFIELD1 (A2)       2 FILLER 2X       2 #RFIELD2 (A2)       2 FILLER 4X       2 #RFIELD3 (A2) END-DEFINE ... </pre>



## 42 ハンドルの定義

---

- 機能 ..... 242
- 構文説明 ..... 242

## ハンドルの定義

DEFINE DATA LOCAL、DEFINE DATA OBJECT、DEFINE DATA PARAMETER、DEFINE DATA INDEPENDENT、および DEFINE DATA CONTEXT で使用する *handle-definition* の構文は、次のとおりです。

<i>handle-name</i>	HANDLE OF OBJECT [	{	CONSTANT	}	<i>init-definition</i>			
			INIT					
	( <i>array-definition</i> ) HANDLE OF	{	OBJECT	}	[[	CONSTANT	]	<i>array-init-definition</i>
					INIT			

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

ハンドルは、コードのダイアログエレメントを特定し、ハンドル変数に格納されます。詳細については、『プログラミングガイド』の「NaturalX」セクションを参照してください。

DEFINE DATA ステートメントでの HANDLE 定義は、ダイアログエレメントまたはダイアログを作成すると、自動的に生成されます。

ハンドルを定義すると、任意のステートメントでハンドル名を使用して、定義したダイアログエレメントタイプの属性値をクエリ、設定、または変更することができます。

### handle-definition の例：

```
1 #SAVEAS-MENUITEM HANDLE OF MENUITEM
1 #OK-BUTTON (1:10) HANDLE OF PUSHBUTTON
```

## 構文説明

<i>handle-name</i>	ハンドルに割り当てる名前です。ユーザー定義変数の命名規則が適用されます。『Natural の使用』ドキュメントの「 <a href="#">ユーザー定義変数の命名規則</a> 」を参照してください。
HANDLE OF OBJECT	『プログラミングガイド』の「NaturalX」セクションで説明されているように、NaturalX と一緒に使用します。
CONSTANT	変数/配列は、指定された定数として処理されます。割り当てられた定数値は、変数/配列が参照されるたびに使用されます。割り当てられた値は、プログラムの実行中には変更できません。  <b>注意:</b>

	<p>1. 内部処理の理由から、1つのグループ定義内で変数定義と定数定義を混在させることは許可されていません。つまり、1つのグループに含めることができるのは、変数か定数のいずれかのみです。</p> <p>2. CONSTANT 節を <code>DEFINE DATA INDEPENDENT</code> および <code>DEFINE DATA CONTEXT</code> で使用することはできません。</p>
<b>INIT</b>	<p>変数／配列に初期値が割り当てられます。この値は、変数／配列が <code>RESET INITIAL</code> ステートメントで参照されるときにも使用されます。</p> <p><b>注意:</b> <code>DEFINE DATA INDEPENDENT</code> および <code>DEFINE DATA CONTEXT</code> では、INIT 節は、その節を含む、各実行プログラミングオブジェクトで評価されます（変数を割当ててるプログラミングオブジェクト内だけではありません）。これは、グローバル変数の INIT の動作とは異なります。</p>
<i>init-definition</i>	<p><i>init-definition</i> オプションを使用して、変数の初期値および定数値を定義します。「<a href="#">初期値の定義</a>」を参照してください。</p>
<i>array-definition</i>	<p><i>array-definition</i> の場合、配列定義の次元の下限と上限を定義します。「<a href="#">配列の次元の定義</a>」を参照してください。</p>
<i>array-init-definition</i>	<p>配列に初期値が割り当てられます。この値は、配列が <code>RESET INITIAL</code> ステートメントで参照されるときにも使用されます。</p>



# 43 配列の次元の定義

---

- 機能 ..... 246
- 構文説明 ..... 246

## 配列の次元の定義

---

*array-dimension-definition* は次のステートメントで使用します。DEFINE DATA OBJECT、および DEFINE DATA LOCAL、DEFINE DATA INDEPENDENT、DEFINE DATA CONTEXT、DEFINE DATA OBJECT の *variable-definition* オプション。

*array-dimension-definition* の構文は、次のとおりです。

```
{bound[:bound]},... 3
```

## 機能

---

*array-dimension-definition* の場合、配列定義の次元の下限と上限を定義します。

最大3次元までの配列を定義することができます。

『プログラミングガイド』の「配列」も参照してください。

## 構文説明

---

<i>bound</i>	<p>次のいずれかが境界になります。</p> <ul style="list-style-type: none"><li>■ 整数定数</li><li>■ 事前に定義された名前付き定数</li><li>■ (データベース配列用に) 事前に定義されたユーザー定義変数</li><li>■ アスタリスク (*) は、拡張可能な境界を定義します。それ以外の場合は、X-array として認識されます。</li></ul> <p>1つの境界だけを指定した場合、その値は上限を表し、下限は1であると想定されます。</p>
--------------	---

### X-Array

配列の少なくとも1つの次元で少なくとも1つの境界を拡張可能に指定すると、その配列は X-array (eXtensible array) と呼ばれます。1つの次元で拡張可能なのは1つの境界だけであり (上限または下限のいずれか)、両方を拡張可能にすることはできません。多次元配列では、例えば #a(1:100, 1:\*) のように、定数と拡張可能な境界が混在していることがあります。



例：

```
DEFINE DATA LOCAL
1 #ARRAY1(I4/1:10)
1 #ARRAY2(I4/10)
1 #X-ARRAY3(I4/1:*)
1 #X-ARRAY4(I4/*,1:5)
1 #X-ARRAY5(I4/*:10)
1 #X-ARRAY6(I4/1:10,100:*,*:1000)
END-DEFINE
```

次の表に、上記のプログラムで使用する配列の境界をわかりやすく示します。

	次元 1		次元 2		次元 3	
	下限	上限	下限	上限	下限	上限
#ARRAY1	1	10	-	-	-	-
#ARRAY2	1	10	-	-	-	-
#X-ARRAY3	1	拡張可能	-	-	-	-
#X-ARRAY4	1	拡張可能	1	5	-	-
#X-ARRAY5	拡張可能	10	-	-	-	-
#X-ARRAY6	1	10	100	拡張可能	拡張可能	1000

配列定義の例：

```
#ARRAY2(I4/10) /* a one-dimensional array with 10 occurrences (1:10)
#X-ARRAY4(I4/*,1:5) /* a two-dimensional array
#X-ARRAY6(I4/1:10,100:*,*:1000) /* a three-dimensional array
```

パラメータデータエリア内の可変配列

パラメータデータエリアには、可変オカレンス数を含む配列を指定できます。この指定は、インデックス表記 1:V を使用して行います。

例 1: #ARR01 (A5/1:V)

例 2: #ARR02 (I2/1:V,1:V)

変数インデックス表記 1:V を含むパラメータ配列は、次の長さでのみ再定義できます。

- 例えば、次のように 1:V インデックスが右端の場合は、そのエレメンタリフィールドの長さになります。

#ARR(A6/1:V) は、最大 6 バイトまでの長さで再定義できます。

#ARR(A6/1:2,1:V) は、最大 6 バイトまでの長さで再定義できます。

#ARR(A6/1:2,1:3,1:V) は、最大 6 バイトまでの長さで再定義できます。

## 配列の次元の定義

- 右端の固定オカレンス数とエレメンタリフィールドの長さの積は、例えば次のようになります。

#ARR(A6/1:V,1:2) は、最大  $2 \times 6 = 12$  バイトまでの長さで再定義できます。

#ARR(A6/1:V,1:3,1:2) は、最大  $3 \times 2 \times 6 = 36$  バイトまでの長さで再定義できます。

#ARR(A6/1:2,1:V,1:3) は、最大  $3 \times 6 = 18$  バイトまでの長さで再定義できます。

変数インデックス表記 1:V は、再定義の中では使用できません。

例：

```
DEFINE DATA PARAMETER
  1 #ARR(A6/1:V)
  1 REDEFINE #ARR
    2 #R-ARR(A1/1:V) /* (1:V) is not allowed in a REDEFINE block
END-DEFINE
```

オカレンス数はコンパイル時には不明であるため、ステートメント **INPUT**、**WRITE**、**READ WORK FILE**、および **WRITE WORK FILE** でインデックス表記 (\*) を使用してオカレンス数を参照することはできません。インデックス表記 (\*) は、すべての次元に適用されるか、またはどの次元にも適用されないかのいずれかになります。

有効な例：

```
#ARR01 (*)
#ARR02 (*,*)
#ARR01 (1)
#ARR02 (5,#FIELDX)
#ARR02 (1,1:3)
```

無効な例：

```
#ARRAYY (1,*) /* not allowed
```

ラインタイムエラーを回避するために、このような配列の最大オカレンス数は、別のパラメータを経由してサブプログラムまたはサブルーチンに渡す必要があります。代わりに、システム変数 \*OCCURRENCE を使用することもできます。

### 注意:

1. インデックス 1:V を含むパラメータデータエリアを使用する場合（つまり、**DEFINE DATA LOCAL** ステートメントで指定した場合）は、V という名前の変数を **CONSTANT** として定義する必要があります。
2. ダイアログでは、インデックス 1:V を **BY VALUE** と一緒に使用することはできません。

# 44 初期値の定義

---

■ 機能 .....	250
■ 制限事項 .....	250
■ 構文説明 .....	250

## 初期値の定義

---

DEFINE DATA LOCAL、DEFINE DATA INDEPENDENT、DEFINE DATA CONTEXT、および DEFINE DATA OBJECT の *variable-definition* オプションで使用する *init-definition* の構文は、次のとおりです。


```
{ <constant>
  <system-variable>
  FULL LENGTH <character-s>
  LENGTH n <character-s> }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

---

*init-definition* オプションを使用して、変数の初期値および定数値を定義します。

 **注意:** *variable-definition* オプションで、キーワード `INIT` が初期化に使用されている場合、変数の内容に影響するステートメントによって値を変更することができます。キーワード `CONST` が初期化に使用されている場合、値を変更しようとすると、コンパイラで拒否されます。

『プログラミングガイド』の「フィールドの定義」の「初期値」も参照してください。

## 制限事項

---

再定義されたフィールドでは、*init-definition* は許可されていません。

## 構文説明

---

<code>&lt;constant&gt;</code>	変数の初期化に使用する定数値、またはフィールドに割り当てる定数値です。定数の詳細については、『プログラミングガイド』の「ユーザ定義定数」を参照してください。
-------------------------------	--

<system-variable>	<p>変数の初期値として Natural システム変数の値を指定することもできます。例：</p> <pre>DEFINE DATA LOCAL 1 #MYDATE (D) INIT &lt;*DATX&gt; END-DEFINE</pre> <p><b>注意:</b> <b>RESET INITIAL</b> ステートメント内でシステム変数が参照されると、システム変数は再び評価されます。つまり、プログラムの実行を開始したときにシステム変数に含まれていた値にリセットされるのではなく、RESET INITIAL ステートメントを実行するときに含まれている値にリセットされます。</p>
<p><b>FULL LENGTH</b> &lt;character-s&gt;</p> <p><b>LENGTH n</b> &lt;character-s&gt;</p>	<p>初期値として、特定の 1 文字または文字列を、変数の全体または一部に挿入します。これは、英数字変数（コードページまたは Unicode）変数だけに有効です。</p> <p>FULL LENGTH オプションを使用すると、指定した <i>character</i> または <i>characters</i> がフィールド全体に挿入されます。次の例では、フィールド全体にアスタリスクが挿入されます。</p> <pre>DEFINE DATA LOCAL 1 #FIELD (A25) INIT FULL LENGTH &lt;'*&gt; END-DEFINE</pre> <p>LENGTH <i>n</i> オプションを使用すると、フィールドの先頭 <i>n</i> 桁に、指定した <i>character</i> または <i>characters</i> が挿入されます。<i>n</i> は数値定数にする必要があります。次の例では、フィールドの先頭 4 桁に感嘆符が挿入されます。</p> <pre>DEFINE DATA LOCAL 1 #FIELD (A25) INIT LENGTH 4 &lt;'!&gt; END-DEFINE</pre>



# 45 配列用の初期値／定数値

---

■ 機能 .....	254
■ 制限事項 .....	254
■ 構文説明 .....	255


DEFINE DATA LOCAL、DEFINE DATA INDEPENDENT、DEFINE DATA CONTEXT、および DEFINE DATA OBJECT の *variable-definition* オプションで使用する *array-init-definition* オプションの構文は、次のとおりです。

```
ALL
{ { { { { index[:index] } ...3 } } } } }
{ { { { { V } } } } }
{ { { { { FULL LENGTH } } } } } <character-s
  { { { { { LENGTH n } } } } } } >
{ { { { { constant } } } } } }
{ { { { { system-variable,... } } } } } } >
...
}
```

## 機能

---

*array-init-definition* を使用して、配列の初期値および定数値を定義します。

 **注意:** *variable-definition* オプションで、キーワード **INIT** が初期化に使用されている場合、変数の内容に影響するステートメントによって値を変更することができます。キーワード **CONST** が初期化に使用されている場合、値を変更しようとする、コンパイラで拒否されます。

『プログラミングガイド』の「フィールドの定義」の「初期値」、および特に次のセクションも参照してください。

- 初期値
- ユーザー定義定数

## 制限事項

---

再定義されたフィールドでは、*array-init-definition* は許可されていません。



## 構文説明

<b>ALL</b>	配列のすべての次元にあるすべてのオカレンスを、同じ値で初期化します。
<i>index</i>	<i>index</i> で指定した配列オカレンスのみを初期化します。 <i>index</i> を指定する場合、 <i>constant</i> で指定できる値は1つだけです。つまり、指定したオカレンスはすべて、同じ値で初期化されます。
<b>V</b>	1つの次元のオカレンスを異なる値で初期化する場合、この表記は多次元配列にのみ関係しています。V は、V で指定した次元のすべてのオカレンスで構成されるインデックス範囲を示します。つまり、その次元のオカレンスは、すべて初期化されます。配列ごとに1つの次元のみを、V で指定することができます。オカレンスは、オカレンスごとにその次元用に指定した値で初期化されます。値の数は、V で指定した次元のオカレンス数を超えないようにする必要があります。
<i>constant</i>	配列を初期化に使用する定数（値）（INIT）、または配列に割り当てる定数（CONSTANT）です。定数の詳細については、『プログラミングガイド』の「ユーザー定義定数」を参照してください。  <b>注意:</b> 値の指定がないオカレンスは、 <b>デフォルト値</b> で初期化されます。
<i>system-variable</i>	配列の初期値として Natural システム変数の値を指定することもできます。  <b>注意:</b> 複数の定数値またはシステム変数を指定する場合は、それぞれを、INPUT 区切り文字（IDセッションパラメータで指定した文字）またはコンマ（,）で区切る必要があります。セッションパラメータ DC でコンマを小数点文字として定義している場合は、この目的でコンマを使用することはできません。
<b>FULL LENGTH</b>  <b>LENGTH <i>n</i></b>	初期値として、特定の1文字または文字列を、変数の全体または一部に挿入します（フォーマット A または U の変数にのみ有効）。  FULL LENGTH オプションを使用すると、指定した <i>character</i> または <i>characters</i> が配列オカレンス全体に挿入されます。  LENGTH <i>n</i> オプションを使用すると、配列オカレンスの先頭 <i>n</i> 桁に、指定した <i>character</i> または <i>characters</i> が挿入されます。  FULL LENGTH または LENGTH <i>n</i> では、 <i>system-variable</i> を指定することはできません。  1つの <i>array-init-definition</i> の中では、FULL LENGTH または LENGTH <i>n</i> のいずれか一方だけを指定できます。両方の表記を混在させることはできません。

配列に対する **LENGTH**  $n$  の例：

この例では、配列の各オカレンスの先頭 5 桁に NONON が挿入されます。

```
DEFINE DATA LOCAL
1 #FIELD (A25/1:3) INIT ALL LENGTH 5 <'NO'>
...
END-DEFINE
```

配列への初期値の割り当てについては、多数の例が『プログラミングガイド』に記載されています。

## 46 フィールド／変数の EM、HD、PM パラメータ

---

- 機能 ..... 258
- 構文説明 ..... 258


DEFINE DATA LOCAL と DEFINE DATA OBJECT の *view-definition* オプション、DEFINE DATA LOCAL、DEFINE DATA INDEPENDENT、DEFINE DATA CONTEXT、および DEFINE DATA OBJECT の *variable-definition* オプションで使用する *emhdpm* オプションの構文は、次のとおりです。

```
((EM=value) [HD='text'] [PM=value])
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

このオプションを使用して、フィールド／変数に有効な追加のパラメータを定義できます。

 **注意:** 指定したデータベースフィールドが編集マスク (EM=) またはヘッダー (HD=) のいずれでもない場合は、**DDM** で指定されているデフォルト編集マスクとデフォルトヘッダーが使用されます。ただし、デフォルト編集マスクとデフォルトヘッダーのいずれか 1 つを指定した場合は、DDM からのもう 1 つのデフォルトは使用されません。

## 構文説明

EM= <i>value</i>	このパラメータは、I/O ステートメントでフィールドを表示する際に使用される編集マスクを定義するために使用します。『パラメータリファレンス』でセッションパラメータ EM を参照してください。
HD=' <i>text</i> '	このパラメータでは、フィールドのデフォルトヘッダーとして使用されるヘッダーを定義します。『パラメータリファレンス』でセッションパラメータ HD を参照してください。
PM= <i>value</i>	このパラメータでは、フィールドの出力方法を示す出力モードを設定します。『パラメータリファレンス』でセッションパラメータ PM を参照してください。

# 47 DEFINE DATA ステートメントの使用例

---

■ 例 1 - DEFINE DATA LOCAL (ダイレクトデータ定義) .....	260
■ 例 2 - DEFINE DATA LOCAL (配列の定義/初期化) .....	260
■ 例 3 - DEFINE DATA (ビューの定義、配列の再定義) .....	261
■ 例 4 - DEFINE DATA (グローバル、パラメータ、およびローカルデータエリア) .....	262
■ 例 5 - DEFINE DATA (初期化) .....	263
■ 例 6 - DEFINE DATA (可変配列) .....	264

次のトピックについて説明します。

## 例 1 - DEFINE DATA LOCAL (ダイレクトデータ定義)

```

** Example 'DDAEX1': DEFINE DATA
*****
DEFINE DATA LOCAL
1 #VAR1      (A15)
1 #VAR2
  2 #VAR2A   (N4.1) INIT <1111>
  2 #VAR2B   (N6.2) INIT <222222>
1 REDEFINE #VAR2
  2 #VAR2C   (A2)
  2 #VAR2D   (A2)
  2 #VAR2E   (A6)
END-DEFINE
*
WRITE NOTITLE '=' #VAR2A / '=' #VAR2B /
              '=' #VAR2C / '=' #VAR2D / '=' #VAR2E
*
END
    
```

プログラム **DDAEX1** の出力：

```

#VAR2A:  1111.0
#VAR2B:  222222.00
#VAR2C:  11
#VAR2D:  11
#VAR2E:  022222
    
```

## 例 2 - DEFINE DATA LOCAL (配列の定義／初期化)

```

** Example 'DDAEX2': DEFINE DATA (array definition/initialization)
*****
DEFINE DATA LOCAL
1 #VAR1 (A1/1:2,1:2) INIT (1,V)  <'A','B'>
1 #VAR2 (N5/1:2,1:3) INIT (1,2)  <200>
1 #VAR3 (A1/1:4,1:3) INIT (V,2:3) <'W','X','Y','Z'>
END-DEFINE
*
WRITE NOTITLE '=' #VAR1 (1,1) '=' #VAR1 (1,2)
              / '=' #VAR1 (2,1) '=' #VAR1 (2,2)
*
WRITE      /// '=' #VAR2 (1,1) '=' #VAR2 (1,2)
              / '=' #VAR2 (2,1) '=' #VAR2 (2,2)
    
```

```

*
WRITE      /// '=' #VAR3 (1,1) '=' #VAR3 (1,2) '=' #VAR3 (1,3)
WRITE      /  '=' #VAR3 (2,1) '=' #VAR3 (2,2) '=' #VAR3 (2,3)
WRITE      /  '=' #VAR3 (3,1) '=' #VAR3 (3,2) '=' #VAR3 (3,3)
WRITE      /  '=' #VAR3 (4,1) '=' #VAR3 (4,2) '=' #VAR3 (4,3)
*
END

```

プログラム **DDAEX2** の出力：

```

#VAR1: A #VAR1: B
#VAR1:  #VAR1:

#VAR2:      0 #VAR2:      200
#VAR2:      0 #VAR2:      0

#VAR3:  #VAR3: W #VAR3: W
#VAR3:  #VAR3: X #VAR3: X
#VAR3:  #VAR3: Y #VAR3: Y
#VAR3:  #VAR3: Z #VAR3: Z

```

### 例 3 - DEFINE DATA (ビューの定義、配列の再定義)

```

** Example 'DDAEX3': DEFINE DATA (view definition, array redefinition)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 ADDRESS-LINE (A20/2)
  2 PHONE
*
1 #ARRAY      (A75/1:4)
1 REDEFINE #ARRAY
  2 #ALINE    (A25/1:4,1:3)
1 #X          (N2) INIT <1>
1 #Y          (N2) INIT <1>
END-DEFINE
*
FORMAT PS=20
LIMIT 5
FIND EMPLOY-VIEW WITH NAME = 'SMITH'

```

```

MOVE NAME          TO #ALINE (#X,#Y)
MOVE ADDRESS-LINE(1) TO #ALINE (#X+1,#Y)
MOVE ADDRESS-LINE(2) TO #ALINE (#X+2,#Y)
MOVE PHONE         TO #ALINE (#X+3,#Y)
IF #Y = 3
  RESET INITIAL #Y
  PERFORM PRINT
ELSE
  ADD 1 TO #Y
END-IF
AT END OF DATA
  PERFORM PRINT
END-ENDDATA
END-FIND
*
DEFINE SUBROUTINE PRINT
  WRITE NOTITLE (AD=OI) #ARRAY(*)
  RESET #ARRAY(*)
  SKIP 1
END-SUBROUTINE
*
END

```

プログラム **DDAEX3** の出力：

```

SMITH          SMITH          SMITH
ENGLANDSVEJ 222 3152 SHETLAND ROAD 14100 ESWORTHY RD.
554349         MILWAUKEE     MONTERREY
              877-4563      994-2260

SMITH          SMITH
5 HAWTHORN     13002 NEW ARDEN COUR
OAK BROOK     SILVER SPRING
150-9351      639-8963

```

## 例4-DEFINE DATA（グローバル、パラメータ、およびローカルデータエリア）

```

** Example 'DDAEX4': DEFINE DATA (global and local data area definition)
*****
DEFINE DATA
GLOBAL
  USING DDAEX4G
LOCAL
1 #FIELD1 (A10)
1 #FIELD2 (N5)
END-DEFINE

```



```

*
MOVE 'HELLO' TO #FIELD1
MOVE 123     TO #FIELD2
*
CALLNAT 'DDAEX4N' #FIELD1 #FIELD2
*
END

```

プログラム **DDAEX4** によって使用されるグローバルデータエリア **DDAEX4G** :

```

1 GLOBAL-FIELD                A    10

```

プログラム **DDAEX4** によって呼び出されるサブプログラム **DDAEX4N** :

```

** Example 'DDAEX4N': DEFINE DATA PARAMETER (called by DDAEX4)
*****
DEFINE DATA
PARAMETER
1 #FIELDA (A10)
1 #FIELDB (N5)
END-DEFINE
*
WRITE '=' #FIELDA '=' #FIELDB
END

```

プログラム **DDAEX4** の出力 :

```

Page      1                                05-01-12  08:55:53
#FIELDA: HELLO      #FIELDB:      123

```

## 例 5 - DEFINE DATA (初期化)

```

** Example 'DDAEX5': DEFINE DATA (initialization)
*****
DEFINE DATA LOCAL
1 #START-DATE (D)   INIT <*DATX>
1 #UNDERLINE   (A50) INIT FULL LENGTH <'_'>
1 #SCALE       (A65) INIT LENGTH 65 <'....+..../'>
END-DEFINE
*
WRITE NOTITLE #START-DATE (DF=L)
              / #UNDERLINE
              / #SCALE
END

```

プログラム **DDAEX5** の出力：

2005-01-12

.....+...../.....+...../.....+...../.....+...../.....+...../.....+...../.....+

## 例 6 - DEFINE DATA (可変配列)

```
** Example 'DDAEX6': DEFINE DATA (variable array with (1:V))
*****
DEFINE DATA LOCAL
1 #ARRAY (A1/1:10)
1 #MAX-ARR (P3)
END-DEFINE
*
#ARRAY (1) := 'R'
#ARRAY (2) := 'E'
#ARRAY (3) := 'D'
#MAX-ARR := 4
*
WRITE #ARRAY(*)
*
CALLNAT 'DDAEX6N' #ARRAY(1:4) #MAX-ARR
*
WRITE #ARRAY(*)
*
*
#MAX-ARR := 5
*
CALLNAT 'DDAEX6N' #ARRAY(1:5) #MAX-ARR
*
WRITE #ARRAY(*)
*
END
```

プログラム **DDAEX6** によって呼び出されるサブプログラム **DDAEX6N**：

```
** Example 'DDAEX6N': DEFINE DATA (variable array with (1:V))
*****
DEFINE DATA
PARAMETER
1 #STRING (A1/1:V)
1 #MAX (P3)
END-DEFINE
*
IF #MAX = 4
MOVE 'B' TO #STRING (1)
MOVE 'L' TO #STRING (2)
```

```
MOVE 'U' TO #STRING (3)
MOVE 'E' TO #STRING (4)
END-IF
*
IF #MAX = 5
MOVE 'W' TO #STRING (1)
MOVE 'H' TO #STRING (2)
MOVE 'I' TO #STRING (3)
MOVE 'T' TO #STRING (4)
MOVE 'E' TO #STRING (5)
END-IF
END
```

プログラム **DDAEX4** の出力：

```
Page      1                                05-01-12  09:06:43
R E D
B L U E
W H I T E
```



# 48 DEFINE PRINTER

---

▪ 機能 .....	268
▪ 構文説明 .....	269
▪ z/OS バッチ、TSO、およびサーバーでのプリンタ名 .....	272
▪ z/VSE バッチ環境でのプリンタ名 .....	275
▪ VM/CMS 環境でのプリンタ名 .....	276
▪ BS2000/OSD バッチおよび TIAM 環境でのプリンタ名 .....	277
▪ CICS 環境でのプリンタ名 .....	283
▪ Com-plete 環境でのプリンタ名 .....	283
▪ Com-plete/SMARTS 環境でのプリンタ名 .....	283
▪ Natural Advanced Facilities 環境でのプリンタ名 .....	284
▪ 追加レポートのプリンタ名 .....	284
▪ 例 .....	285

```

DEFINE PRINTER ([logical-printer-name]=]n)
    [OUTPUT operand1]
    {
        PROFILE operand2
        CODEPAGE operand2
        FORMS operand2
        NAMEoperand2
        DISPOoperand2
        CLASSoperand2
        COPIESoperand3
        PRTYoperand4
    }
    ...7
    
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT END OF PAGE](#) | [AT TOP OF PAGE](#) | [CLOSE PRINTER](#) | [DISPLAY](#) | [EJECT](#) | [FORMAT](#) | [NEWPAGE](#) | [PRINT](#) | [SKIP](#) | [SUSPEND IDENTICAL SUPPRESS](#) | [WRITE](#) | [WRITE TITLE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

DEFINE PRINTER ステートメントは、記号名をレポート番号に割り当て、論理出力先へのレポートの割り当てを制御するために使用します。これにより、さまざまな論理印刷キューへの出力をより柔軟に作成できます。

このステートメントの実行時に、指定したプリンタがすでに開いていた場合、プリンタはステートメントにより暗黙的に閉じられます。ただし、プリンタを明示的に閉じる場合は、[CLOSE PRINTER](#) ステートメントを使用する必要があります。

DEFINE PRINTER ステートメントの詳細については、「[Natural プログラミング言語での Unicode およびコードページのサポート](#)」セクションの「ステートメント」を参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S	A U	可	不可
<i>operand2</i>	C S	A U	可	不可
<i>operand3</i>	C S	N	可	不可
<i>operand4</i>	C S	N P I	可	不可

構文要素の説明：

<i>(n)</i>	<p>プリンタ番号：</p> <p>セッションの間に使用するすべての出力ファイルは、プロファイルパラメータ PRINT のサブパラメータ AM (アクセスメソッドのタイプ) によってアクセスメソッドに事前に割り当てるか、または JCL での定義 (AM=STD のみ) によって自動的に割り当てられます。</p> <p>プリンタ番号 <i>n</i> は、0~31 の範囲にある値になります。この番号は、<b>DISPLAY/WRITE</b> または <b>CLOSE PRINTER</b> ステートメントでも使用されます。</p> <p>プリンタ番号 0 はハードコピープリンタを示します。一部のアクセスメソッドでは、AM=PC などのハードコピープリンタはサポートされていません。</p>
<i>logical-printer-name</i>	<p>論理プリンタ名：</p> <p>オプションで、論理名 <i>logical-printer-name</i> をプリンタ <i>n</i> に割り当てることができます。この名前は、<b>DISPLAY/WRITE</b> ステートメントで <i>rep</i> 表記に使用できます。</p> <p><i>logical-printer-name</i> の命名規則はユーザー定義変数と同じです。同じプリンタ番号に複数の論理名を割り当てることができます。OUTPUT オペランド (下記参照) の値とは異なり、<i>logical-printer-name</i> はコンパイル時に評価されるので、プログラム制御フローから独立しています。</p>
<b>OUTPUT</b> <i>operand1</i>	<p>プリンタ名：</p> <p><i>operand1</i> では、オンラインスプーリングシステム内でプリンタ名を指定したり、プリンタ番号に割り当てる出力ファイル名や追加プリンタの名前を指定したりすることができます。詳細については、下記の「<b>追加レポートのプリンタ名</b>」を参照してください。プロファイルパラメータ PRINT のサブパラメータ DEST で、最初に 8 バイトの論理プリンタ名を定義できます。そのデフォルト値はアクセスメソッドのタイプによって異なり、また <i>operand1</i> で上書きされることもあります。</p>

	<p><i>operand1</i>の長さは1～253文字です。<i>operand1</i>が変数の場合は、その長さを少なくとも8バイトにする必要があります。名前には、プリンタまたは論理/物理データセット名のいずれかを指定できます。可能なフォーマットは、オペレーティングシステム環境、およびこのプリンタ番号に対してプロファイルパラメータ PRINT のサブパラメータ AM で定義するアクセスメソッドによって異なります。</p> <p>指定した名前がすでに別のプリンタ番号に対して定義されており、さらにこのプリンタが未使用（つまり閉じた状態）の場合、指定したプリンタ番号に対してプロファイルパラメータ PRINT のサブパラメータ ROUTE=ON を設定していると、出力はこのプリンタにルーティングされます。<i>operand1</i>と一致するプリンタ名がない場合は、最も大きな番号を持つ未使用のプリンタが使用され、またその名前は <i>operand1</i> で上書きされます。出力のルーティングは、SYSFILE コマンドを使用してユーザーに表示することはできません。</p> <p>各オペレーティングシステムまたは各 TP モニタにおけるプリンタの命名規則については、次のセクションを参照してください。</p> <ul style="list-style-type: none"> <li>■ <b>z/OS</b> バッチ、<b>TSO</b>、およびサーバーでのプリンタ名</li> <li>■ <b>z/VSE</b> バッチ環境でのプリンタ名</li> <li>■ <b>VM/CMS</b> 環境でのプリンタ名</li> <li>■ <b>BS2000/OSD</b> バッチおよび <b>TIAM</b> 環境でのプリンタ名</li> <li>■ <b>CICS</b> 環境でのプリンタ名</li> <li>■ <b>Com-plete</b> 環境でのプリンタ名</li> <li>■ <b>Com-plete/SMARTS</b> 環境でのプリンタ名</li> <li>■ <b>Natural Advanced Facilities</b> 環境でのプリンタ名</li> <li>■ 追加レポートのプリンタ名</li> </ul> <p>下記の節を使用すると、TP モニタまたはオペレーティングシステムのスプリーティングシステムによって解釈される印刷制御情報を提供できます。これらの節を1つまたは複数指定することができますが、それぞれの節の指定は1回のみです。</p>
<p><b>PROFILE</b> <i>operand2</i></p>	<p>プリンタ制御文字テーブルの名前：</p> <p>PROFILE 節では、<i>operand2</i>としてプリンタ制御文字テーブルの名前を指定できます。<i>operand2</i>の最大長は8バイトです。</p> <p>プリンタ制御文字テーブルは、プロファイルパラメータ CCTAB（プリンタエスケープシーケンス定義）で定義できます。</p> <p><b>注意:</b> Natural Advanced Facilities を使用すると、プリンタ制御文字テーブルのメンテナンスをオンラインで行うことができます（『Natural Advanced Facilities』ドキュメントを参照）。</p>
<p><b>CODEPAGE</b> <i>operand2</i></p>	<p>コードページの名前：</p> <p>CODEPAGE では、NATCONFIG モジュールで指定されているコードページの名前（フォーマット/長さ：A64）を示します。</p>



	CODEPAGE を各 OUTPUT の出力先に適用しない場合、CODEPAGE は無視されます。
--	---

## スプーリングシステムのパラメータ

下記の節を使用すると、TP モニタのスプーリングシステムのパラメータに値を指定できます。これらの節のデフォルト値は、プロファイルパラメータ PRINT の対応するサブパラメータで設定できます（*DEFINE PRINTER* ステートメントのキーワードサブパラメータの説明を参照）。

プリンタを閉じると、すべてのオプションがデフォルト値にリセットされます。Natural 環境での定義が不明な場合は、各モジュールで、*DEFINE PRINTER* ステートメントを使用して定義を設定することをお勧めします。

<b>FORMS</b> <i>operand2</i>	<p>フォーム：</p> <p>オペランドの最大長は 8 バイトです。この節のデフォルト値は、プロファイルパラメータ PRINT のサブパラメータ FORMS で設定できます。</p>
<b>NAME</b> <i>operand2</i>	<p>リスト名：</p> <p>オペランドの最大長は 8 バイトです。この節のデフォルト値は、プロファイルパラメータ PRINT のサブパラメータ NAME で設定できます。</p>
<b>DISP</b> <i>operand2</i>	<p>出力先：</p> <p>オペランドの最大長は 4 バイトです。</p> <p>DISP 節で指定できる <i>operand2</i> の値は、DEL、HOLD、KEEP、および LEAV です。この節のデフォルト値は、プロファイルパラメータ PRINT のサブパラメータ DISP で設定できます。DISP 節を省略した場合（または指定が正しくない場合）は、デフォルトで DEL が適用されます。</p>
<b>CLASS</b> <i>operand2</i>	<p>スプールクラス：</p> <p>オペランドの最大長は 1 バイトです。この節のデフォルト値は、プロファイルパラメータ PRINT のサブパラメータ CLASS で設定できます。</p>
<b>COPIES</b> <i>operand3</i>	<p>コピー部数：</p> <p><i>operand3</i> は整数値でなければなりません。この節のデフォルト値は、プロファイルパラメータ PRINT のサブパラメータ COPIES で設定できます。</p>
<b>PRTY</b> <i>operand4</i>	<p>リスト表示のプロパティ：</p> <p>設定可能値：1～255。 <i>operand4</i> は整数値でなければなりません。この節のデフォルト値は、プロファイルパラメータ PRINT のサブパラメータ PRTY で設定できます。</p>

## z/OS バッチ、TSO、およびサーバーでのプリンタ名

このセクションでは、次のトピックについて説明します。

- 論理データセット名
- 物理データセット名
- HFS ファイル
- JES スプールファイルクラス
- NULLFILE
- データセットの割り当てと割り当て解除
- サーバー環境での出力ファイル

アクセスメソッド AM=STD を指定してプリンタ番号を定義する場合は、*operand1* を使用して、そのプリンタ番号に割り当てられる論理または物理データセット名を指定できます。

*operand1* は 1~253 文字で指定し、また次のいずれかになります。

- 論理データセット名 (DD 名、1~8 文字)。
- カタログデータセットの物理データセット名 (1~44 文字)、または物理データセット名 (データセット名の 1~44 文字、それに加えてメンバ名用のカッコの中に 1~8 文字)。
- MVS UNIX Services 環境の HFS ファイル (1~253 文字) のパス名およびメンバ名。
- JES スプールファイルクラス。
- NULLFILE (ダミーのデータセットを示すため)

### 論理データセット名

例：

```
DEFINE PRINTER (21) OUTPUT 'SYSPRINT'
```

DD 名 SYSPRINT で指定したデータセットは、DEFINE PRINTER ステートメントを実行する前に割り当てる必要があります。詳細については、後述の「[データセットの割り当てと割り当て解除](#)」を参照してください。

割り当ては、JCL、CLIST (TSO)、またはダイナミックアロケーション (SVC 99) によって実行できます。ダイナミックアロケーションには、ライブラリ SYSEXT にあるアプリケーションプログラミングインターフェイス USR2021N を使用できます。

DEFINE PRINTER ステートメントで定義したデータセット名は、プロファイルパラメータ PRINT のサブパラメータ DEST で指定した名前よりも優先されます。

オプションで、データセット名に接頭辞 DDN= を付けて、それが DD 名であることを示すことができ、また追加レポートを使用して名前の競合を回避することもできます。次に例を示します。

```
DEFINE PRINTER (22) OUTPUT 'DDN=SOURCE'
```

## 物理データセット名

例：

```
DEFINE PRINTER (23) OUTPUT 'TEST.PRINT.FILE'
```

指定したデータセットが、カタログフォーム内に存在する必要があります。DEFINE PRINTER ステートメントを実行すると、現在の DD 名とオプション `DISP=SHR` を含む SVC 99 によって、データセットはダイナミックに割り当てられます。詳細については、後述の「[データセットの割り当てと割り当て解除](#)」を参照してください。

データセット名が 8 文字以下でピリオド (.) を含んでいない場合は、誤って DD 名と解釈されている可能性があります。これを回避するには、名前の先頭に DSN= を付けます。次に例を示します。

```
DEFINE PRINTER (22) OUTPUT 'DSN=PRINTXYZ'
```

データセットが PDS メンバの場合は、データセット名 (1~44 文字) の後に PDS メンバ名 (1~8 文字) をカッコで囲んで指定します。次に例を示します。

```
DEFINE PRINTER (4) OUTPUT 'TEST.PRINT.PDS(TEST1)'
```

指定したメンバが存在しない場合は、指定した名前の新しいメンバが作成されます。

## HFS ファイル

例：

```
DEFINE PRINTER (14) OUTPUT '/u/nat/rec/test.txt'
```

指定したパス名が存在する必要があります。DEFINE PRINTER ステートメントを実行すると、HFS ファイルがダイナミックに割り当てられます。指定したメンバが存在しない場合は、指定した名前の新しいメンバが作成されます。

データセットのダイナミックアロケーションでは、次の z/OS パスオプションを使用します。

```
PATHOPTS=(OCREAT,OTRUNC,ORDWR)
PATHMODE=(SIRUSR,SIWUSR,SIRGRP,SIWGRP)
FILEDATA=TEXT
```

HFS ファイルを閉じると、z/OS によってそのファイルの割り当てが自動的に解除されます（プロファイルパラメータ PRINT のサブパラメータ FREE の設定に関係なく）。

## JES スプールファイルクラス

JES スプールデータセットを作成するには、SYSOUT=x（x は目的のスプールファイルクラス）を指定します。デフォルトスプールファイルクラスの場合は、SYSOUT=\* を指定します。

例：

```
DEFINE PRINTER (10) OUTPUT 'SYSOUT=A'
DEFINE PRINTER (12) OUTPUT 'SYSOUT=*'
```

ダイナミックアロケーションの追加パラメータを指定するには、DEFINE PRINTER ステートメントの代わりに、ライブラリ SYSEXT にあるアプリケーションプログラミングインターフェイス USR2021N を使用します。

## NULLFILE

ダミーのデータセットを割り当てるには、*operand1* として NULLFILE を指定します。

```
DEFINE PRINTER (n) OUTPUT 'NULLFILE'
```

これは、次の JCL 定義に対応しています。

```
// DD-name DD DUMMY
```

## データセットの割り当てと割り当て解除

DEFINE PRINTER ステートメントを実行するときに、物理データセット名、HFS ファイル、スプールファイルクラス、またはダミーデータセットがすでに指定してある場合、対応するデータセットはダイナミックに割り当てられます。論理出力ファイルがすでに開いていると、その論理出力ファイルは自動的に閉じられます。ただし、プロファイルパラメータ PRINT のサブパラメータ CLOSE=FIN（閉じるタイミング）が指定してある場合は、エラーが発行されます。さらに、新しいデータセットを割り当てる前に、同じ現在の DD 名で割り当てられている既存のデータセットは、自動的に割り当てが解除されます。

プログラムの開始時にまだ割り当てられていない出力ファイルのオープンに失敗して不要なオーバーヘッドが発生するのを防止するために、プロファイルパラメータ PRINT のサブパラメータ OPEN=ACC（最初のアクセス時に開く）を指定して出力ファイルを定義する必要があります。

HFS ファイルの場合、またはプロファイルパラメータ PRINT のサブパラメータ FREE=ON を指定して出力ファイルを定義した場合は、これらのファイルを閉じると、直ちに出力ファイルの割り当てが自動的に解除されます。

データセットのダイナミックな割り当てと割り当て解除の代わりとして、ライブラリ SYSEXT でアプリケーションプログラミングインターフェイス USR2021N が提供されています。また、この API を使用すると、ダイナミックアロケーションの追加パラメータを指定することもできます。

## サーバー環境での出力ファイル

サーバー環境では、複数の Natural セッションによって同じ DD 名でデータセットの割り当てまたはオープンが試みられると、エラーが発生することがあります。これを回避するには、プロファイルパラメータ PRINT のサブパラメータ DEST=\* で出力ファイルを指定するか、または DEFINE PRINTER ステートメントで OUTPUT '\*' を指定して、その出力ファイルに対して最初の DEFINE PRINTER ステートメントを実行する際に、Natural が物理データセットの割り当てで一意的な DD 名を生成するようにします。

DD 名が CM で始まるすべての出力ファイルは、サーバー環境内のすべてのセッションによって共有されます。共有された出力ファイルは最初のセッションによって開かれ、サーバーを終了すると物理的に閉じられます。詳細については、『オペレーション』ドキュメントの「サーバーとしての Natural」セクションを参照してください。

## z/VSE バッチ環境でのプリンタ名

アクセスメソッド AM=STD を指定してプリンタ番号を定義する場合、*operand1* は次のようになります。

- 論理データセット名（DD 名、1~7 文字）
- NULLFILE（ダミーのデータセットを示すため）

## 論理データセット名

例：

```
DEFINE PRINTER (2) OUTPUT 'SYSOUT1'
```

指定するデータセット SYSOUT1 は、JCL や、z/VSE の標準またはパーティションラベルで定義されている必要があります。

DEFINE PRINTER ステートメントで指定したデータセット名は、プロファイルパラメータ PRINT のサブパラメータ DEST で指定した名前よりも優先されます。

オプションで、データセット名に接頭辞 DDN= を付けて、それが DD 名であることを示すこともできます。次に例を示します。

```
DEFINE PRINTER (5) OUTPUT 'DDN=MYPRINT'
```

## NULLFILE

ダミーのデータセットを割り当てるには、*operand1* として NULLFILE を指定します。

```
DEFINE PRINTER (n) OUTPUT 'NULLFILE'
```

## VM/CMS 環境でのプリンタ名

---

アクセスメソッド AM=STD を指定してプリンタ番号を定義する場合は、*operand1* を使用して、そのプリンタ番号に割り当てられる論理または物理データセット名を指定できます。

その際に、z/OS 環境と同じことが適用されますが（「[z/OS バッチ、TSO、およびサーバーでのプリンタ名](#)」を参照）、次の点で異なります。

- MVS SVC 99 によるダイナミックアロケーションの代わりに、CMS コマンド FILEDEF を使用してデータセットを定義します。
- HFS ファイルはサポートされていません。
- JES スプールクラスはサポートされていません。

- また、次の構文を使用します。

```
DEFINE PRINTER (n) OUTPUT ('fname ftype fmode(options)')
```

これは CMS コマンドを生成します。

```
FILEDEF ddname-n DISK fname ftype fmode (options)
```

- さらに、次の構文を使用できます。

```
DEFINE PRINTER (n) OUTPUT ('FILEDEF=filedef-parameters')
```

これは CMS コマンドを生成します。

```
FILEDEF ddname-n =filedef-parameters
```

## BS2000/OSD バッチおよび TIAM 環境でのプリンタ名

アクセスメソッド AM=STD を指定してプリンタ番号を定義する場合は、*operand1* を使用して、そのプリンタ番号に割り当てられるファイル名、リンク名、またはシステムファイルを指定できます。

この場合、*operand1* は長さが 1~253 文字になり、次のいずれかを表します。

- リンク名
- ファイル名
- 一般的なファイル名
- ファイル名とリンク名
- 一般的なファイル名とリンク名
- システムファイル SYSOUT
- システムファイル SYSLST
- システムファイル SYSLSTnn - nn=01~99
- システムファイル SYSLSTnn の暗黙的な割り当て

- \*DUMMY

次の規則が適用されます。

1. ファイル名とリンク名は、位置指定パラメータとしてもキーワードパラメータとしても指定できます。対応するキーワードは FILE= と LINK= です。位置指定パラメータとキーワードパラメータの混在は可能ですが、お勧めしません。

2. コンマなしの1~8文字長の文字列はリンク名と解釈されます。この表記はNaturalの旧バージョンと互換性があります。例：

```
DEFINE PRINTER (1) OUTPUT 'P01'
```

キーワードパラメータを使用した対応する定義は次のとおりです。

```
DEFINE PRINTER (1) OUTPUT 'LINK=P01'
```

3. コンマなしの9~54文字長の文字列はファイル名と解釈されます。例：

```
DEFINE PRINTER (2) OUTPUT 'NATURAL31.TEST.PRINTER02'
```

キーワードパラメータを使用した対応する定義は次のとおりです。

```
DEFINE PRINTER (2) OUTPUT 'FILE=NATURAL31.TEST.PRINTER02'
```

4. 次の入力は、長さを考慮せずに解釈されるので、ルール2と3の例外です。

- キーワード入力：LINK=、FILE=
- \*DUMMY
- NULLFILE (\*DUMMYと同じ)
- \*
- \*,\*
- SYSOUT
- SYSLST または SYSLST(*nn*)

例：DEFINE PRINTER (7) OUTPUT 'FILE=Y' は文字列に含まれる文字数が9文字未満ですが、有効なファイル割り当てであり、リンク名ではありません。

5. 一般的なファイル名の形式は次のとおりです。

```
pnn.userid.tsn.date.time.number
```



ここでは次の内容を表しています。

<i>nn</i>	レポート番号
<i>userid</i>	Natural ユーザー ID、8 文字
<i>tsn</i>	現在のタスクの BS2000/OSD TSN、4 桁
<i>date</i>	DDMMYYYY
<i>time</i>	HHISS
<i>number</i>	シーケンシャル番号、5 桁

6. 一般的なリンク名の形式は次のとおりです。

```
NPFnnnnn
```

*nnnnn* は 5 桁の数字で、ダイナミックリンク名が生成されるたびに 1 ずつ増加します。

7. プリンタ番号へのファイル割り当てを変更すると、割り当てられていた出力ファイルは CLOSE により暗黙的に閉じられます。

リンク名のみを指定する場合（例：P01）を除き、すべての場合においてキーワードパラメータを使用することをお勧めします。これにより、追加レポートとの名前の競合を防ぐことができます。9 文字未満のファイル名を使用する場合には必須です。

例：

```
DEFINE PRINTER (1) OUTPUT 'LINK=SOURCE'
DEFINE PRINTER (1) OUTPUT 'FILE=SOURCE'
DEFINE PRINTER (1) OUTPUT 'SOURCE'
```

## リンク名

例：

```
DEFINE PRINTER (1) OUTPUT 'LINKP01'
```

次の構文と同じ意味です。

```
DEFINE PRINTER (1) OUTPUT 'LINK=LINKP01'
```

LINK LINKP01 を含むファイルが、実行時に存在している必要があります。これは、Natural を開始する前に JCL を使用するか、または現在のアプリケーションからダイナミックに割り当てることにより、作成することができます。ダイナミックアロケーションには、ライブラリにあるアプリケーションプログラミングインターフェイス USR2029 を使用できます。実行前に、リンクが別のファイルへの宛先として有効になっていた場合（例：P01）、プロファイルパラメータ PRINT のサブパラメータ FREE の値（指定可能な値は ON および OFF）に応じて、リンクは開放または保持されます。解放は、BS2000/OSD コマンドプロセッサに対する明示的な RELEASE コールによって実行されます。

## ファイル名

例：

```
DEFINE PRINTER (2) OUTPUT 'NATURAL31.TEST.PRINTER02'
```

次の構文と同じ意味です。

```
DEFINE PRINTER (2) OUTPUT 'FILE=NATURAL31.TEST.PRINTER02'
```

*operand1* で指定したファイルは、FILE マクロコールを使用して設定され、また DEFINE PRINTER ステートメントの実行前に対応する出力ファイルに対して有効だったリンク名を継承します。

## 一般的なファイル名

例：

```
DEFINE PRINTER (21) OUTPUT '*'
```

次の構文と同じ意味です。

```
DEFINE PRINTER (21) OUTPUT 'FILE=*'
```

ルール 4 に従って作成した名前を持つファイルは、FILE マクロコールを使用して設定され、また DEFINE PRINTER ステートメントの実行前に対応する出力ファイルに対して有効だったリンク名を継承します。

```
DEFINE PRINTER (22) OUTPUT 'FILE=*,LINK=GENFLK22'
```

ルール 4 に従って作成した名前を持つファイルには、指定したリンク名が FILE マクロコールを使用して設定されます。

## ファイル名とリンク名

例：

```
DEFINE PRINTER (11) OUTPUT 'NATURAL31.TEST.PRINTER11,LNKP11'
```

次の構文と同じ意味です。

```
DEFINE PRINTER (11) OUTPUT 'FILE=NATURAL31.TEST.PRINTER11,LINK=LNKP11'
```

次の構文も同じ意味です。

```
DEFINE PRINTER (11) OUTPUT 'FILE=NATURAL31.TEST.PRINTER11,LNKP11'
```

*operand1* で指定したファイルには、指定したリンク名が FILE マクロコールを使用して設定されます。このファイルは、対応するプリンタ番号に割り当てられます。

## 一般的なファイル名とリンク名


例：

```
DEFINE PRINTER (27) OUTPUT '*,*'
```

次の構文と同じ意味です。

```
DEFINE PRINTER (27) OUTPUT 'FILE=*,LINK=*'
```

ルール 4 およびルール 5 に従って作成したファイル名を持つファイルは、FILE マクロコールを使用して設定され、指定したプリンタ番号 (27) に割り当てられます。

 **注意:** ファイル名とリンク名を指定した場合、プロファイルパラメータ PRINT のサブパラメータ FREE の値に関係なく、以前のリンク名は解放されません。

## システムファイル **SYSOUT**

例：

```
DEFINE PRINTER (14) OUTPUT 'SYSOUT'
```

レポート 14 を SYSOUT に書き込みます。

TIAM 環境：デフォルトでは、SYSOUT は画面への出力です。

## システムファイル **SYSLST**

例：

```
DEFINE PRINTER (15) OUTPUT 'SYSLST'
```

レポート 15 をシステムファイル SYSLST に書き込みます。

## システムファイル **SYSLSTnn - nn=01~99**

例：

```
DEFINE PRINTER (16) OUTPUT 'SYSLST16'
```

レポート 16 をシステムファイル SYSLST16 に書き込みます。

## システムファイル **SYSLSTnn** の暗黙的な割り当て

例：

```
DEFINE PRINTER (11) OUTPUT 'SYSLST=LST.PRINTER11'
```

システムファイル SYSLST をファイル LST.PRINTER11 に割り当て、レポート 11 をシステムファイル SYSLST に書き込みます。

```
DEFINE PRINTER (13) OUTPUT 'SYSLST13=LST.PRINTER13'
```

システムファイル SYSLST13 をファイル LST.PRINTER13 に割り当て、レポート 13 をシステムファイル SYSLST13 に書き込みます。

```
DEFINE PRINTER (19) OUTPUT 'SYSLST19=*'
```

システムファイル SYSLST19 をルール 4 に従って生成した名前のファイルに割り当て、レポート 19 をシステムファイル SYSLST19 に書き込みます。

## CICS 環境でのプリンタ名

アクセスメソッド AM=CICS を指定してプリンタ番号を定義する場合は、プリンタのプロファイルパラメータ PRINT のサブパラメータ TYPE に合わせて、*operand1* を一時データまたは一時ストレージキューの名前（1～8 文字）にすることができます。TYPE=TD（一時データ）では、*operand1* の最初の 4 文字のみが保持されるため、事前に一時データの送り先として CICS を指定しておく必要があります。

詳細については、「CICS 環境での *Natural* の出力ファイルおよびワークファイル」（『TP モニタインターフェイス』ドキュメント）も参照してください。

## Com-plete 環境でのプリンタ名

AM=COMP を指定すると、有効なプリンタ番号（TID）または論理プリンタ名を割り当てることができます。次に例を示します。

```
DEFINE PRINTER (1) OUTPUT '11'
DEFINE PRINTER (2) OUTPUT 'P102'
```

## Com-plete/SMARTS 環境でのプリンタ名

AM=SMARTS を指定すると、任意のプリンタ名を割り当てることができます。次に例を示します。

```
DEFINE PRINTER (14) OUTPUT '/nat/path/printer'
DEFINE PRINTER (14) OUTPUT '/nat/path/printer/file/'
DEFINE PRINTER (14) OUTPUT 'printer'
```

ファイルが SMARTS ポータブルファイルシステムまたはネイティブファイルシステムのどちらにあるかは、SMARTS の MOUNT\_FS パラメータによって異なります。パスの最初の要素 (/nat/) により、ターゲットファイルシステムが決定されます。

文字列の最後がスラッシュ (/) の場合、最後の要素が出力ファイルの名前になります。それ以外の場合は、ユーザー ID とシーケンス番号からファイルの名前が生成されます。文字列の先頭がスラッシュでない場合は、ファイルのパスが環境変数 \$NAT\_PRINT\_ROOT から取得されます。

指定したパス名が存在する必要があります。DEFINE PRINTER ステートメントを実行すると、HFS ファイルがダイナミックに割り当てられます。指定したメンバが存在しない場合は、指定した名前の新しいメンバが作成されます。

## Natural Advanced Facilities 環境でのプリンタ名

Natural Advanced Facilities ユーザーに対しては、事前に定義した任意の論理プリンタプロファイルの名前を指定することができます。この論理プリンタプロファイルは、現在有効になっているユーザープロファイルに属している必要はありません。論理プリンタプロファイルは、NATSPool ファイルで定義した論理プリンタプロファイルである場合もあります。論理プリンタプロファイルは、DEFINE PRINTER ステートメントを含む Natural プログラムが実行されている間のみ有効になります。詳細については、『Natural Advanced Facilities』ドキュメントを参照してください。

## 追加レポートのプリンタ名

追加レポートには、デフォルトで次の名前が割り当てられます。

レポート	機能
BROADCAST	TP モニタ端末へのメッセージ行出力。Com-plete 環境以外では、MESSAGE（下記を参照）と同じ機能です。目的の端末ユニット上でアクティブになっているトランザクションがなくなるまで、メッセージはその端末に送信されません。
CCONTROL	CCONTROL は $n-1$ に関連付けられている特別なプリンタ制御テーブルの名前であり、変更はできません。詳細については、「Printer-Advance 制御文字」（『オペレーション』ドキュメント）を参照してください。
CONNECT	Connect フォルダ内への出力。  Natural のインストールに関する注意：Natural の NATPCNT モジュールを、Natural ニュークリアスにリンクする必要があります。
DUMMY	削除する出力。
HARDCOPY	現在のハードコピーデバイスへの出力。
INCORE	NSPF インコアデータベース内への出力。
INFOLINE	Natural 情報行への出力。情報行の詳細については、Natural 端末コマンド %X を参照してください。
MESSAGE	TP モニタ端末へのメッセージ行出力。メッセージの先頭の 8 バイトには、対象の端末 ID を格納する必要があります。TSO および CMS では、端末 ID の代わりにユーザー ID

レポート	機能
	が必要になります。ライブラリ SYSEXTP では、例としてプログラム MSGSW が提供されています。
SOURCE	Natural ソースエリアへの出力。
WORKPOOL	Natural ISPF ワークプール内への出力。

## 例

- [例 1 - Com-plete でのプリンタ名の定義](#)
- [例 2 - バッチ環境でのプリンタ名の定義](#)
- [例 3 - 情報行への出力](#)
- [例 4 - 事前に定義されたプリンタを含むセッションの使用](#)

### 例 1 - Com-plete でのプリンタ名の定義

```
/* PRINTER NAME DEFINITION FOR COM-PLETE
*
DEFINE PRINTER (1) OUTPUT 'TID100'
WRITE (1) 'PRINTED ON PRINTER TID100'
END
```

### 例 2 - バッチ環境でのプリンタ名の定義

```
/* OUTPUT ON 'SYSPRINT' (FOR BATCH ENVIRONMENTS)
*
DEFINE PRINTER (REPORT1 = 1) OUTPUT 'SYSPRINT'
WRITE (REPORT1) 'REPORT 1 PRINTED ON PRINTER SYSPRINT'
*
/* OUTPUT TO DEFAULT PRINTER DESTINATION
/* DEFINED WITH PROFILE PARAMETER 'PRINT', SUBPARAMETER 'DEST'
*
DEFINE PRINTER (REPORT2 = 2)
WRITE (REPORT2) 'REPORT PRINTED TO DESTINATION'
```

### 例 3 - 情報行への出力

```

** Example 'DPIEX1': DEFINE PRINTER
*****
*
SET CONTROL 'XI+'          /* SWITCH INFOLINE MODE ON
SET CONTROL 'XT'          /* INFOLINE TOP
*
DEFINE PRINTER (1) OUTPUT 'INFOLINE'
WRITE (1) 'EXECUTING' *PROGRAM 'BY' *INIT-USER
WRITE 'TEST OUTPUT'
EJECT                     /* FORCE PHYSICAL I/O
*
SET CONTROL 'X'           /* SWITCH BACK TO NORMAL
*
END

```

プログラム **DPIEX1** の出力：

```

EXECUTING DPIEX1   BY HTR
Page      1                               05-01-13  14:54:33
TEST OUTPUT

```

### 例 4 - 事前に定義されたプリンタを含むセッションの使用

```

** Example 'DPREX1': DEFINE PRINTER
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
END-DEFINE
*
* USE SESSION WITH DEFINED PRINTER 1
*
DEFINE PRINTER (INVOICE-LIST=1) OUTPUT 'OUTQ1'
LIMIT 5
READ EMPL-VIEW BY NAME
  WRITE (INVOICE-LIST) NAME
END-READ
*
END

```



# 49

## DEFINE SUBROUTINE

---

▪ 機能 .....	288
▪ 制限事項 .....	288
▪ 構文説明 .....	289
▪ サブルーチンで使用可能なデータ .....	290
▪ 例 .....	290

```
DEFINE [SUBROUTINE] subroutine-name
    statement ...
{
    END-SUBROUTINE
    RETURN (レポーティングモードのみ)
}
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CALL](#) | [CALL FILE](#) | [CALL LOOP](#) | [CALLNAT](#) | [ESCAPE](#) | [FETCH](#) | [PERFORM](#)

関連機能グループ：「[プログラムおよびルーチンの呼び出し](#)」

## 機能

---

DEFINE SUBROUTINE ステートメントでは、Natural サブルーチンを定義します。サブルーチンは [PERFORM](#) ステートメントによって呼び出されます。

### 内部／外部サブルーチン

サブルーチンは、[PERFORM](#) ステートメントを含むオブジェクトの中でも（[内部サブルーチン](#)）、[PERFORM](#) を含むオブジェクトの外でも（[外部サブルーチン](#)）定義できます。内部サブルーチンは、それを参照する最初の [PERFORM](#) ステートメントの前/後いずれにでも定義できます。



**注意:** プログラム機能を複数の外部サブルーチンで構築すると、プログラム構造が明確になります。ただし、外部サブルーチンの呼び出しは、内部コードまたは内部サブルーチンに比べてオーバーヘッドが多くなるため、常に外部サブルーチンにはできるだけ大きな機能ブロックを含めるようにすることをお勧めします。

## 制限事項

---

- サブルーチン内で開始した処理ループは、すべて [END-SUBROUTINE](#) を発行する前に閉じる必要があります。
- 内部サブルーチンの中で、別の [DEFINE SUBROUTINE](#) ステートメントを指定することはできません（下記の「[例1](#)」を参照）。
- 外部サブルーチン（サブルーチンタイプのオブジェクト）に、複数の [DEFINE SUBROUTINE](#) ステートメントブロックを指定することはできません（下記の「[例2](#)」を参照）。ただし、外部の [DEFINE SUBROUTINE](#) ブロックの中に内部サブルーチンを指定することはできます（下記の「[例1](#)」を参照）。

例 1

次のような構造は、タイプがサブルーチンのオブジェクトでは有効ですが、その他のオブジェクト（SUBR01 が内部サブルーチンとみなされます）の場合は無効です。

```

...
DEFINE SUBROUTINE SUBR01
  ...
  PERFORM SUBR02
  PERFORM SUBR03
  ...
  DEFINE SUBROUTINE SUBR02
  /* inline subroutine...
  END-SUBROUTINE
  ...
  DEFINE SUBROUTINE SUBR03
  /* inline subroutine...
  END-SUBROUTINE
END-SUBROUTINE
END

```

例 2（無効）

次の構造は、オブジェクトタイプがサブルーチンの場合には無効です。

```

...
DEFINE SUBROUTINE SUBR01
...
END-SUBROUTINE
DEFINE SUBROUTINE SUBR02
...
END-SUBROUTINE
END

```

構文説明

<i>subroutine-name</i>	サブルーチン名（最大 32 文字）の命名規則は、ユーザー定義変数名の規則と同じです。『Natural の使用』ドキュメントの「ユーザー定義変数の命名規則」を参照してください。  サブルーチン名は、該当のサブルーチンが定義されているモジュールの名前に依存しません（同じ名前にする事は可能ですが、同じ名前にする必要はありません）。
<b>END-SUBROUTINE</b>	サブルーチン定義は END-SUBROUTINE で終了します。
<b>RETURN</b>	レポートモードでは、RETURN でもサブルーチンを終了できます。

## サブルーチンで使用可能なデータ

このセクションでは、次のトピックについて説明します。

- インラインサブルーチン
- 外部サブルーチン

### インラインサブルーチン

明示的なパラメータを、呼び出し元プログラムから **PERFORM** ステートメントを介して内部サブルーチンに渡すことはできません。

内部サブルーチンからは、呼び出し元のプログラム内で定義してあるローカルデータエリアおよびグローバルデータエリアのデータにアクセスできます。

### 外部サブルーチン

外部サブルーチンは、現在設定されているグローバルデータエリアにアクセスできます。また、**PERFORM** ステートメントで、呼び出し元のオブジェクトから外部サブルーチンにパラメータを直接渡せるため、グローバルデータエリアのサイズを縮小できます。

外部サブルーチンでは、呼び出し元のプログラムで定義したローカルデータエリアは使用できませんが、独自のローカルデータエリアを持つことはできます。

## 例

- 例 1 - サブルーチンの定義
- 例 2 - GDA フィールドを使用する外部サブルーチンのサンプル構造

### 例 1 - サブルーチンの定義

```
** Example 'DSREX1S': DEFINE SUBROUTINE (structured mode)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 ADDRESS-LINE (A20/2)
  2 PHONE
*
1 #ARRAY (A75/1:4)
1 REDEFINE #ARRAY
  2 #ALINE (A25/1:4,1:3)
1 #X (N2) INIT <1>
```

```

1 #Y      (N2) INIT <1>
END-DEFINE
*
FORMAT PS=20
LIMIT 5
FIND EMPLOY-VIEW WITH NAME = 'SMITH'
  MOVE NAME           TO #ALINE (#X,#Y)
  MOVE ADDRESS-LINE(1) TO #ALINE (#X+1,#Y)
  MOVE ADDRESS-LINE(2) TO #ALINE (#X+2,#Y)
  MOVE PHONE          TO #ALINE (#X+3,#Y)
  IF #Y = 3
    RESET INITIAL #Y
    PERFORM PRINT
  ELSE
    ADD 1 TO #Y
  END-IF
  AT END OF DATA
  PERFORM PRINT
END-ENDDATA
END-FIND
*
DEFINE SUBROUTINE PRINT
  WRITE NOTITLE (AD=OI) #ARRAY(*)
  RESET #ARRAY(*)
  SKIP 1
END-SUBROUTINE
*
END

```

プログラム **DSREX1S** の出力：

SMITH ENGLANDSVEJ 222 554349	SMITH 3152 SHETLAND ROAD MILWAUKEE 877-4563	SMITH 14100 ESWORTHY RD. MONTERREY 994-2260
SMITH 5 HAWTHORN OAK BROOK 150-9351	SMITH 13002 NEW ARDEN COUR SILVER SPRING 639-8963	

レポートモードの例については、ライブラリ SYSEXRM のプログラム **DSREX1R** を参照してください。

## 例 2 - GDA フィールドを使用する外部サブルーチンのサンプル構造

```

** Example 'DSREX2': DEFINE SUBROUTINE (using GDA fields)
*****
DEFINE DATA
GLOBAL
  USING DSREX2G
END-DEFINE
*
INPUT 'Enter value in GDA field' GDA-FIELD1
*
* Call external subroutine in DSREX2S
*
PERFORM DSREX2-SUB
*
END

```

プログラム **DSREX2** によって使用されるグローバルデータエリア **DSREX2G** :

```

1 GDA-FIELD1                A      2

```

プログラム **DSREX2** によって呼び出されるサブルーチン **DSREX2S** :

```

** Example 'DSREX2S': SUBROUTINE (external subroutine using global data)
*****
DEFINE DATA
GLOBAL
  USING DSREX2G
END-DEFINE
*
DEFINE SUBROUTINE DSREX2-SUB
*
  WRITE 'IN SUBROUTINE' *PROGRAM '=' GDA-FIELD1
*
END-SUBROUTINE
*
END

```

# 50

## DEFINE WINDOW

---

▪ 機能 .....	294
▪ 構文説明 .....	295
▪ ウィンドウの入力フィールドの保護 .....	299
▪ 他のウィンドウの呼び出し .....	299
▪ 例 .....	299

```

DEFINE WINDOW window-name
[
  SIZE {
    AUTO
    QUARTER
    operand1* operand2
  }
]
[
  BASE {
    CURSOR
    {
      TOP
      BOTTOM
    }
    {
      LEFT
      RIGHT
    }
    operand3/ operand4
  }
]
[REVERSED [(CD=background-color)]]
[TITLEoperand5]
[
  CONTROL {
    WINDOW
    SCREEN
  }
]
[
  FRAMED {
    [ON] [(CD=frame-color)] [position-clause]
    OFF
  }
]

```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[DEFINE WINDOW](#) | [INPUT](#) | [REINPUT](#) | [SET WINDOW](#)

関連機能グループ：「[対話型処理用の画面生成](#)」

## 機能

DEFINE WINDOW ステートメントでは、ウィンドウのサイズ、位置、属性を指定します。

ウィンドウとは、端末画面上に表示される、プログラムによって構築された論理ページのセグメントのことです。存在を認識できなくても、常にウィンドウは存在しています。別のウィンドウが指定されない限り、ウィンドウのサイズは端末画面の物理サイズと同じです。

DEFINE WINDOW ステートメントはウィンドウをアクティブ化しません。ウィンドウをアクティブ化するには、[SET WINDOW](#) ステートメントを使用するか、INPUT ステートメントの [WINDOW](#) 節を使用します。

『プログラミングガイド』の「[アプリケーションユーザーインターフェイスの設計](#)」にある「[画面設計](#)」セクションの「[ウィンドウ](#)」も参照してください。



**注意:** Natural ウィンドウ、つまり最新のウィンドウは常に1つです。画面に前のウィンドウが表示されていても、アクティブではないので、Natural に無視されます。現在のウィンドウにのみ入力できます。入力する十分なスペースがない場合は、まずウィンドウサイズを調整する必要があります。



## フルスクリーンの制御

ウィンドウがアクティブな場合でも、Naturalによってフルスクリーンの制御が行われます。これは CICS や TSO のようなシングルセッションシステムに影響を与えません。Natural が Complete や Multi-pass のようなマルチプルセッションシステムの環境下で動作している場合、中断された Natural セッションが再開すると、完全な Natural 画面（つまり現在アクティブなウィンドウだけでなく、「その下」の Natural 画面も含む）が表示されます。同時に、フルスクリーンのフィールド属性はウィンドウによって部分的にオーバーレイされますが、ウィンドウによる影響はありません。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S	N P I	可	不可
<i>operand2</i>	C S	N P I	可	不可
<i>operand3</i>	C S	N P I	可	不可
<i>operand4</i>	C S	N P I	可	不可
<i>operand5</i>	C S	A U	可	不可

構文要素の説明：


<i>window-name</i>	<i>window-name</i> では、ウィンドウの名前を指定します。名前の最大長は 32 文字です。ウィンドウ名には、ユーザー定義変数と同じ命名規則が適用されます。 『Natural の使用』ドキュメントの「ユーザー定義変数の命名規則」を参照してください。
<b>SIZE</b>	SIZE 節では、ウィンドウのサイズを指定します。  <b>注意:</b> メインフレーム環境では、Natural でデータを画面上に表示できるようにするには属性バイトと呼ばれる追加の列が必要です（他のプラットフォームでは、そのような属性バイトは不要）。したがって、メインフレーム環境では他のプラットフォームよりもウィンドウによってオーバーレイされる画面領域が大きく、ウィンドウ内に表示できるページセグメントのサイズが小さくなります。  例：ウィンドウサイズを SIZE 5 * 15（15 桁の幅）として定義します。  ■ メインフレーム環境では、ウィンドウによってオーバーレイされる画面領域は 16 桁で、ウィンドウ内に表示できるサイズはフレームなしで 14 桁、フレーム付きで 10 桁です。  ■ 他のプラットフォームでは、ウィンドウによってオーバーレイされる画面領域は 15 桁で、ウィンドウ内に表示できるサイズはフレームなしで 15 桁、フレーム付きで 13 桁です。

<p><b>SIZE AUTO</b></p>	<p>ウィンドウのサイズは、Natural 実行時に自動で決定されます。サイズは、ウィンドウに生成されるデータによって、次のように決定されます。</p> <ul style="list-style-type: none"> <li>■ ウィンドウの行数は、生成される INPUT 行の数です (PF キー行、メッセージ行、および情報/統計行を加えることが可能)。</li> <li>■ ウィンドウの桁数は、最も長い INPUT 行で決まります。Natural では、行の終わりから行の有効バイトの右端までが検索されます。これにより、入力だけのフィールド (AD=A) または修正可能フィールド (AD=M) は、切り捨てられる可能性があります。これを回避するには、このようなフィールドの後に単一文字を置くか、または次の節でウィンドウサイズを明示的に設定します。</li> </ul> <p style="text-align: center;">SIZEoperand1 * operand2</p> <p>SIZE 節を省略すると、デフォルトで SIZE AUTO が適用されます。</p> <p><b>注意:</b> タイトルはウィンドウデータの一部ではありません。したがって、ウィンドウサイズが上記のように指定され、さらにタイトルがウィンドウよりも長い場合、タイトルは切り捨てられます。</p>
<p><b>SIZE QUARTER</b></p>	<p>ウィンドウサイズは物理画面の 1/4 になります。</p>
<p><b>SIZE operand1 * operand2</b></p>	<p>ウィンドウサイズは <math>n</math> 行 <math>\times</math> <math>n</math> 桁になります。operand1 では行数、operand2 では桁数を指定します。2 つのオペランドのいずれにも小数桁を含めることはできません。</p> <p><b>FRAMED</b> を指定している場合、指定サイズにはフレームも含まれます。</p> <p>有効な最小ウィンドウサイズは、次のようになります。</p> <ul style="list-style-type: none"> <li>■ フレームなしの場合は 2 行 <math>\times</math> 10 桁</li> <li>■ フレーム付きの場合は 4 行 <math>\times</math> 13 桁</li> </ul> <p>有効な最大ウィンドウサイズは、物理画面のサイズです。</p>
<p><b>BASE</b></p>	<p>BASE 節では、物理画面上でのウィンドウ位置を決定します。BASE 節を省略すると、デフォルトで BASE CURSOR が適用されます。</p>
<p><b>BASE CURSOR</b></p>	<p>ウィンドウの左上隅を、現在のカーソル位置に配置します。カーソル位置とは、画面上の物理的なカーソルの位置です。ウィンドウサイズが原因でウィンドウをカーソル位置に配置できない場合、Natural では、目的の位置にできるだけ近くなるようにウィンドウが自動的に配置されます。</p>
<p><b>BASE TOP/BOTTOM LEFT/RIGHT</b></p>	<p>それぞれ、物理画面上の左上隅、左下隅、右上隅、右下隅にウィンドウを配置します。</p>
<p><b>BASE operand3/operand4</b></p>	<p>物理画面の指定した行/桁に、ウィンドウの左上隅を配置します。operand3 では行数、operand4 では桁数を指定します。2 つのオペランドのいずれにも小数桁を含めることはできません。</p> <p>ウィンドウサイズが原因で指定の位置にウィンドウを配置できない場合は、エラーメッセージが返されます。</p>
<p><b>REVERSED</b></p>	<p>REVERSED を指定すると、ウィンドウが反転表示されます。ただし、この指定は、使用画面でこの機能がサポートされている場合にのみ有効となります。サポートされていない場合、REVERSED は無視されます。</p>

<p><b>REVERSED CD=</b> <i>background-color</i></p>	<p>ウィンドウを反転表示し、さらにウィンドウの背景を指定の色で表示します。ただし、この指定は、使用画面でこれらの機能がサポートされている場合にのみ有効となります。サポートされていない場合、各指定は無視されます。</p> <p>有効なカラーコードについては、『パラメータリファレンス』でセッションパラメータ CD を参照してください。</p>
<p><b>TITLE</b> <i>operand5</i></p>	<p>TITLE 節では、ウィンドウの見出しを指定できます。指定したタイトル (<i>operand5</i>) は、ウィンドウの先頭フレーム行の中央に表示されます。タイトルは、テキスト文字列 (アポストロフで囲む)、またはユーザー定義変数の内容として指定できます。タイトルがウィンドウよりも長い場合、タイトルは切り捨てられます。タイトルは、<b>FRAMED</b> を指定した場合にのみ表示されます。<b>FRAMED OFF</b> を指定すると、TITLE 節は無視されます。</p> <p><b>注意:</b> タイトルの末尾にある空白は削除されます。タイトルの先頭にある文字が空白の場合は、自動的に 1 桁の空白がタイトルに付加されます。</p>
<p><b>CONTROL</b></p>	<p>CONTROL 節を使用して、PF キー行、メッセージ行、および統計行をウィンドウに表示するかフル物理画面に表示するかを指定します。</p>
<p><b>CONTROL WINDOW</b></p>	<p>CONTROL WINDOW を指定すると、ウィンドウ内に PF キー行、メッセージ行、および統計行が表示されます。CONTROL 節を省略すると、デフォルトで CONTROL WINDOW が適用されます。</p>
<p><b>CONTROL SCREEN</b></p>	<p>CONTROL SCREEN を指定すると、ウィンドウの外側にある物理画面全体に PF キー行、メッセージ行、および統計行が表示されます。</p>
<p><b>FRAMED</b></p>	<p>デフォルトでは、つまり FRAMED 節を省略した場合は、ウィンドウにフレームが付きます。</p> <p>最上部と最下部のフレーム行はカーソルに依存します。適用可能なところで、適切な記号 (&lt;, -, +, &gt;: 下記の <i>position-clause</i> を参照) の上にカーソルを置いて Enter キーを押すと、ウィンドウ内でページを前後左右に移動できます。記号が表示されない場合は、カーソルを枠線の最上部 (前のページに戻る場合) または最下部 (次のページに進む場合) に置いて Enter キーを押すことにより、ウィンドウ内でページを前後に移動することができます。</p> <p><b>注意:</b> ウィンドウサイズが 4 行 × 12 (メインフレーム環境では 13) 桁より小さい場合、フレームは表示されません。</p>
<p><b>FRAMED OFF</b></p>	<p>FRAMED OFF を指定すると、フレーム表示および各フレーム関連の指定 (ウィンドウタイトルおよび位置情報) はオフになります。</p>
<p><b>FRAMED (CD=</b> <i>frame-color)</i></p>	<p>この節では、ウィンドウに表示するフレームの色を指定します。ただし、使用画面がカラー端末でない場合、カラー指定は無視されます。</p> <p>有効なカラーコードについては、『パラメータリファレンス』でセッションパラメータ CD を参照してください。</p> <p><b>注意:</b> Natural for Windows では、この指定は無視されます。</p>
<p><i>position-clause</i></p>	<p>POSITION 節は、メインフレームのコンピュータだけで評価されます。他のすべてのプラットフォームでは無視されます。詳細については、下記の「<b>POSITION 節</b>」を参照してください。</p>

## POSITION 節

POSITION 節は、メインフレームのコンピュータだけで評価されます。他のすべてのプラットフォームでは無視されます。



POSITION 節を指定すると、論理ページ上でのウィンドウの位置に関する情報がウィンドウのフレーム内に表示されます。これは、論理ページがウィンドウよりも大きい場合にのみ適用されます。そうでない場合は、POSITION 節は無視されます。位置情報は、論理ページが拡張する方向（現在のウィンドウの上、下、左、右）を示します。

POSITION 節を省略すると、デフォルトで POSITION SYMBOL TOP RIGHT が適用されます。

<b>POSITION SYMBOL</b>	位置情報を記号形式 More: < - + > で表示します。情報は、上部か下部のいずれか、または両方のフレームラインに表示されます。
<b>TOP/BOTTOM</b>	位置情報を上のフレーム行または下のフレーム行のどちらに表示するかを決定します。
<b>AUTO</b>	論理ページの横のサイズがウィンドウ内に収まる場合、つまり、マイナス記号文字 (-) またはプラス記号文字 (+) が表示される場合にのみ適用されます。この場合、AUTO によって、記号はそれぞれ文字列 Top、Bottom、More に自動で切り替えられます。
<b>SHORT</b>	記号 < - + > の前にある文字列 More: を非表示にします。
<b>LEFT/RIGHT</b>	位置情報をフレーム行の左または右のどちらに表示するかを決定します。
<b>POSITION TEXT</b>	位置情報をテキスト形式で表示します。情報を上か下のフレーム行に More、Top、Bottom の文字列で表示します。このテキストは、言語依存であり、言語コードによって別の言語でも表示できます。
<b>POSITION TEXT MORE</b>	文字列 Top と Bottom を非表示にして、適用される場所（上、下またはその両方のフレーム行）に文字列 More のみを表示します。
<b>LEFT/RIGHT</b>	位置情報をフレーム行の左または右のどちらに表示するかを決定します。
<b>POSITION OFF</b>	位置情報を非表示にします。位置情報は表示されません。

## ウィンドウの入力フィールドの保護

全体をウィンドウ内に収容できない入力フィールド（AD=A または AD=M）には、次の規則が適用されます。

- フィールドの始まりがウィンドウ内でない入力フィールドは常に保護されます。
- ウィンドウ内で始まりウィンドウ外で終わる入力フィールドは、フィールドに含まれる値がウィンドウ内に完全に表示できない場合にのみ保護されます。この場合、フィールド長ではなく値の長さがウィンドウのサイズを超えているかどうかを決定します。充填文字（プロファイルパラメータ FC で指定）は、値の一部には数えられません。

保護された入力フィールドにアクセスする場合は、フィールドの先頭および値の終わりがウィンドウ内に入るようにウィンドウサイズを調整する必要があります。

## 他のウィンドウの呼び出し

DEFINE WINDOW ステートメントは、論理条件ステートメントブロック内では指定できません。条件によって異なるウィンドウを呼び出すには、条件内で別の SET WINDOW ステートメント（または WINDOW 節を含む INPUT ステートメント）を使用します。

## 例

```
** Example 'DWDEX1': DEFINE WINDOW
*****
DEFINE DATA LOCAL
01 #I (P3)
END-DEFINE
*
SET KEY PF1='%W<<' PF2='%W>>' PF4='%W--' PF5='%W++'
*
DEFINE WINDOW WIND1
    SIZE QUARTER
    BASE TOP RIGHT
    FRAMED ON POSITION SYMBOL AUTO
*
SET WINDOW 'WIND1'
FOR #I = 1 TO 10
    WRITE 25X #I 'THIS IS SOME LONG TEXT' #I
END-FOR
*
END
```

プログラム DWDEX1 の出力：

```

+-----More:      + >+
> r                ! Page      1                !
All  ....+....1....+....2....+....3.. !          !
0010 ** Example 'DWDEX1': DEFINE WIND !          1 THIS !
0020 ***** !          2 THIS !
0030 DEFINE DATA LOCAL !          3 THIS !
0040 01 #I (P3) !          4 THIS !
0050 END-DEFINE !          5 THIS !
0060 * !          6 THIS !
0070 SET KEY PF1='%W<<' PF2='%W>>' PF !          7 THIS !
0080 * ! MORE !
0090 DEFINE WINDOW WIND1 +-----+
0100     SIZE QUARTER
0110     BASE TOP RIGHT
0120     FRAMED ON POSITION SYMBOL AUTO
0130 *
0140 SET WINDOW 'WIND1'
0150 FOR #I = 1 TO 10
0160   WRITE 25X #I 'THIS IS SOME LONG TEXT' #I
0170 END-FOR
0180 *
0190 END
0200
      ....+....1....+....2....+....3....+....4....+....5....+... S 19  L 1

```


# 51

## DEFINE WORK FILE

---

▪ 機能 .....	302
▪ 構文説明 .....	302
▪ z/OS バッチ、TSO、およびサーバーでのワークファイル名 .....	304
▪ z/VSE バッチでのワークファイル名 .....	307
▪ VM/CMS でのワークファイル名 .....	308
▪ BS2000/OSD バッチおよび TIAM でのワークファイル名 .....	309
▪ CICS でのワークファイル名 .....	313
▪ Com-plete/SMARTS でのワークファイル名 .....	313

DEFINE WORK FILE *n* [*operand1*] [TYPE *operand2*]

 **注意:** 角カッコ[...]で示されている要素はオプションですが、このステートメントではこれらのオプションの少なくとも1つを指定する必要があります。

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CLOSE WORK FILE](#) | [READ WORK FILE](#) | [WRITE WORK FILE](#)

関連機能グループ：「[ワークファイル/PC ファイルの制御](#)」


## 機能

DEFINE WORK FILE ステートメントは、Natural アプリケーション内で Natural ワークファイル番号にファイル名を割り当てるために使用します。

このステートメントを使用することで、Natural セッション内でワークファイルの割り当てをダイナミックに行ったり、変更することができます。また、別のレベルで作成したワークファイルの割り当てを上書きすることもできます。

このステートメントを実行したときに、指定したワークファイルがすでに開かれていると、ワークファイルはこのステートメントによって暗黙的に閉じられます。

セッションの間に使用するすべてのワークファイルは、プロファイルパラメータ WORK のサブパラメータ AM によってアクセスメソッドに事前に割り当てるか、または JCL での定義によって自動的に割り当てられます。

 **注意:** Unicode およびコードページのサポートについては、『[Unicode およびコードページのサポート](#)』ドキュメントの「[メインフレームプラットフォーム上のワークファイルと出力ファイル](#)」を参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S	A	可	不可
<i>operand2</i>	C S	A	可	不可



構文要素の説明：

<p><b>DEFINE WORK FILE</b> <i>n</i></p>	<p><i>n</i>は、ワークファイル番号です (1~32)。これは、<b>WRITE WORK FILE</b>、<b>READ WORK FILE</b>、または <b>CLOSE WORK FILE</b> ステートメントのいずれかで使用される番号です。</p>	
<p><i>operand1</i></p>	<p><i>operand1</i> は、ワークファイルの名前です。 <i>operand1</i>では、ワークファイル番号に割り当てられるデータセットの名前を指定できます。 <i>operand1</i>の長さは1~253文字です。名前には、論理または物理データセット名のいずれかを指定できます。可能なフォーマットは、オペレーティングシステム環境、およびプロファイルパラメータ <b>WORK</b> のサブパラメータ <b>AM</b> で定義するアクセスメソッドによって異なります。一部のアクセスメソッドでは、<i>operand1</i>のようなワークファイル名はサポートされていません (例：AM=COMP や AM=PC)。</p> <p><i>operand1</i>を指定しない場合は、このワークファイル番号に対して以前に実行した <b>DEFINE WORK FILE</b> ステートメントで指定された現在の名前によって、<i>operand1</i>の値が決定されます。まだ <b>DEFINE WORK FILE</b> ステートメントを実行していない場合、名前は <b>Natural</b> パラメータモジュール <b>NATPARM</b> から取得されます。</p> <p><b>注意:</b> <i>operand1</i>が指定されていない場合、<b>Natural for Mainframes</b> と <b>Natural for Windows/UNIX/OpenVMS</b> の動作は異なります。</p> <p>各オペレーティングシステムまたは各TPモニタにおけるワークファイルの命名規則については、次のセクションを参照してください。</p> <ul style="list-style-type: none"> <li>■ <b>z/OS</b> バッチ、<b>TSO</b>、およびサーバーでのワークファイル名</li> <li>■ <b>z/VSE</b> バッチでのワークファイル名</li> <li>■ <b>VM/CMS</b> でのワークファイル名</li> <li>■ <b>BS2000/OSD</b> バッチおよび <b>TIAM</b> でのワークファイル名</li> <li>■ <b>CICS</b> でのワークファイル名</li> <li>■ <b>Com-plete/SMARTS</b> でのワークファイル名</li> </ul>	
<p><i>operand2</i></p>	<p><i>operand2</i>はワークファイルのタイプを指定します。 <i>operand2</i>の値では、大文字と小文字が区別されません。値は引用符で囲むか、英数字の変数で指定する必要があります。</p> <p>UNFORMATTED</p>	<p>フォーマットがまったく行われていないファイル。フォーマット情報が書き込まれていません (フィールドまたはレコードのいずれにも)。</p> <p>UNFORMATTED は、ワークファイルをレコード境界のないバイトストリームとして扱います。タイプ UNFORMATTED は、<b>Entire Connection</b> によって拒否されます。</p> <p>フォーマット：UNFORMATTED</p>

	FORMATTED	FORMATTEDでは、正規のレコード指向のワークファイルを定義します。このワークファイルは、旧バージョンの Natural と同じ方法で処理されます。
--	-----------	--

例：

```
DEFINE WORK FILE 17 #FILE TYPE 'UNFORMATTED'
#TYPE := 'FORMATTED'
DEFINE WORK FILE 18 #FILE TYPE #TYPE
```

## z/OS バッチ、TSO、およびサーバーでのワークファイル名

次のトピックについて説明します。

- ワークファイル名 - operand1
- データセットの割り当てと割り当て解除
- サーバー環境でのワークファイル

### ワークファイル名 - operand1

z/OS環境で、アクセスメソッド AM=STD を指定してワークファイル番号を定義する場合、*operand1* は次のようになります。

- 論理データセット名 (DD 名、1~8 文字)。
- カタログデータセットの物理データセット名 (1~44 文字) または物理データセットのメンバ名。
- MVS UNIX Services 環境の **HFS** ファイル (1~253 文字) のパス名およびメンバ名。
- **JES** スプールファイルクラス。
- NULLFILE。

論理データ セット名	<p>例：</p> <pre>DEFINE WORK FILE 21 'SYSOUT1'</pre> <p>指定したデータセット SYSOUT1 は、DEFINE WORK FILE ステートメントを実行する前に割り当てておく必要があります。</p> <p>割り当ては、JCL、CLIST (TCO)、またはダイナミックアロケーション (SVC99) によって実行できます。ダイナミックアロケーションには、ライブラリ SYSEXT にあるアプリケーションプログラミングインターフェイス USR2021N を使用できます。</p> <p>DEFINE WORK FILE ステートメントで定義されているデータセット名は、プロファイルパラメータ WORK のサブパラメータ DEST で指定された名前より優先されます。</p>
---------------	---

	<p>または、データセット名に接頭辞 DDN= を付けて、それが DD 名であることを示すこともできます。次に例を示します。</p> <pre>DEFINE WORK FILE 22 'DDN=MYWORK'</pre>
<p><b>物理データ セット名</b></p>	<p>例：</p> <pre>DEFINE WORK FILE 23 'TEST.WORK.FILE'</pre> <p>指定したデータセットが、カタログフォーム内に存在している必要があります。DEFINE WORK FILE ステートメントを実行すると、現在の DD 名とオプション DISP=SHR を含む SVC 99 によって、データセットはダイナミックに割り当てられます。</p> <p>データセット名が 8 文字以下でピリオド (.) を含んでいない場合は、誤って DD 名と解釈されている可能性があります。これを回避するには、名前の先頭に DSN= を付けます。次に例を示します。</p> <pre>DEFINE WORK FILE 22 'DSN=WORKXYZ'</pre> <p>データセットが PDS メンバの場合は、データセット名 (1~44 文字) の後に PDS メンバ名 (1~8 文字) をカッコで囲んで指定します。次に例を示します。</p> <pre>DEFINE WORK FILE 4 'TEST.WORK.PDS(TEST1)'</pre> <p>指定したメンバが存在しない場合は、指定した名前の新しいメンバが作成されます。</p>
<p><b>HFS ファイル</b></p>	<p>例：</p> <pre>DEFINE WORK FILE 14 '/u/nat/rec/test.txt'</pre> <p>指定したパス名が存在する必要があります。DEFINE WORK FILE ステートメントを実行すると、HFS ファイルがダイナミックに割り当てられます。指定したメンバが存在しない場合は、指定した名前の新しいメンバが作成されます。</p> <p>データセットのダイナミックアロケーションでは、次の z/OS パスオプションを使用します。</p> <pre>PATHOPTS=(OCREAT,OTRUNC,ORDWR) PATHMODE=(SIRUSR,SIWUSR,SIRGRP,SIWGRP) FILEDATA=TEXT</pre> <p>HFS ファイルを閉じると、z/OS によってそのファイルの割り当てが自動的に解除されます (プロファイルパラメータ WORK のサブパラメータ FREE の設定に関係なく)。</p> <p>HFS ファイルを読み取るには、OTRUNC オプションの関係で、DEFINE WORK FILE ステートメントの代わりにアプリケーションプログラミングインターフェイス USR2021 (ダイナミックデータセットアロケーション) を使用する必要があります。このオプションでは、HFS ファイルは最初の読み取りアクセス時にリセットされて、空のファイルになります。</p>

<b>JES スプール ファイルクラ ス</b>	<p>JES スプールデータセットを作成するには、SYSOUT=x (xは目的のスプールファイルクラス) を指定します。デフォルトスプールファイルクラスの場合は、SYSOUT=* を指定します。</p> <p>例：</p> <pre>DEFINE WORK FILE 10 'SYSOUT=A' DEFINE WORK FILE 12 'SYSOUT=*'</pre> <p>ダイナミックアロケーションの追加パラメータを指定するには、DEFINE WORK FILE ステートメントの代わりに、ライブラリ SYSEXT にあるアプリケーションプログラミングインターフェイス USR2021N (ダイナミックデータセットアロケーション) を使用します。</p>
<b>NULLFILE</b>	<p>ダミーのデータセットを示します。</p>

### データセットの割り当てと割り当て解除

DEFINE WORK FILE ステートメントを実行するとき、物理データセット名、HFS ファイル、スプールファイルクラス、またはダミーデータセットがすでに指定してある場合、対応するデータセットが自動的に割り当てられます。論理ファイルがすでに開いていると、その論理ファイルは自動的に閉じられます。ただし、プロファイルパラメータ WORK のサブパラメータ CLOSE=FIN を指定してある場合は、エラーが発行されます。また、新しいデータセットを割り当てる前に、同じ現在の DD 名で割り当てられている既存のデータセットは、自動的に割り当てが解除されます。プログラムの開始時にまだ割り当てられていないワークファイルのオープンに失敗して不要なオーバーヘッドが発生するのを防止するために、プロファイルパラメータ WORK のサブパラメータ OPEN=ACC (最初のアクセス時に開く) を指定してワークファイルを定義する必要があります。

HFS ファイルの場合、またはプロファイルパラメータ WORK のサブパラメータ FREE=ON を指定してワークファイルを定義した場合は、これらのファイルを閉じると、直ちにワークファイルの割り当てが自動的に解除されます。

データセットのダイナミックな割り当てと割り当て解除の代わりとして、ライブラリ SYSEXT でアプリケーションプログラミングインターフェイス USR2021N (ダイナミックデータセットアロケーション) が提供されています。また、この API を使用すると、ダイナミックアロケーションの追加パラメータを指定することもできます。

### サーバー環境でのワークファイル

サーバー環境では、複数の Natural セッションによって同じ DD 名でデータセットの割り当てまたはオープンが試みられると、エラーが発生することがあります。これを回避するには、プロファイルパラメータ WORK のサブパラメータ DEST=\* でワークファイルを指定するか、または実際の DEFINE WORK FILE ステートメントの前にプログラム内で DEFINE WORK FILE '\*' を指定して、そのワークファイルに対して最初の DEFINE WORK FILE ステートメントを実行するときに、Natural が物理データセットの割り当てで一意な DD 名を生成するようにします。

DD 名が CM で始まるすべてのワークファイルは、サーバー環境内のすべてのセッションによって共有されます。最初のセッションによって出力用に開かれた共有ワークファイルは、サーバーを終了すると物理的に閉じられます。入力用に開かれた共有ワークファイルが物理的に閉じられるのは、最後のセッションがそのワークファイルを閉じるとき、つまり最後のセッションがエンドオブファイル条件を受信するときです。ワークファイルの読み取りが複数同時に行われた場合は、1つのファイルレコードが1つの `READ WORK FILE` ステートメントだけに提供されます。

## z/VSE バッチでのワークファイル名

z/VSE 環境で、アクセスメソッド `AM=STD` を指定してワークファイル番号を定義する場合、*operand1* は次のようになります。

- 論理データセット名 (DD 名、1~7 文字)
- NULLFILE (ダミーのデータセットを示すため)

論理データセット名	<p>例：</p> <pre>DEFINE WORK FILE 21 'SYSOUT1'</pre> <p>指定するデータセット <code>SYSOUT1</code> は、JCL や、z/VSE の標準またはパーティションラベルで定義されている必要があります。</p> <p><code>DEFINE WORK FILE</code> ステートメントで定義されているデータセット名は、プロファイルパラメータ <code>WORK</code> のサブパラメータ <code>DEST</code> で指定された名前より優先されます。</p> <p>または、データセット名に接頭辞 <code>DDN=</code> を付けて、それが DD 名であることを示すこともできます。次に例を示します。</p> <pre>DEFINE WORK FILE 22 'DDN=MYWORK'</pre>
NULLFILE	<p>ダミーのデータセットを割り当てるには、<i>operand1</i> として <code>NULLFILE</code> を指定します。</p> <pre>DEFINE WORK FILE n 'NULLFILE'</pre>

## VM/CMS でのワークファイル名

VM/CMS 環境で、AM=STD を指定してワークファイルを定義する場合、*operand1* には z/OS 環境と同じことが適用されますが（[上記](#)を参照）、次の点で異なります。

- MVS SVC 99 によるダイナミックアロケーションの代わりに、CMS コマンド FILEDEF を使用してファイルを定義します。
- HFS ファイルはサポートされていません。
- JES スプールクラスはサポートされていません。
- また、次の構文を使用します。

```
DEFINE WORK FILE n 'fname ftype fmode (options)'
```

これは CMS コマンドを生成します。

```
FILEDEF ddname-n DISK fname ftype fmode (options)
```

- さらに、次の構文を使用できます。

```
DEFINE WORK FILE n 'FILEDEF=filedef-parameters'
```

これは CMS コマンドを生成します。

```
FILEDEF ddname-n =filedef-parameters
```

次に例を示します。

```
DEFINE WORK FILE 5 'FILEDEF=TAP1 SL 2 VOLID BKUP08 (BLKSIZE 20000)'
```

これは CMS コマンドを生成します。

```
FILEDEF CMWKF05 TAP1 SL 2 VOLID BKUP08
```

AM=CMS を指定してワークファイルを定義する場合は、DEFINE WORK FILE を使用して送り先を変更できます。パラメータマクロ NETWORK のサブパラメータ DEST で指定できる送り先以外に、REXX のステムも定義できます。詳細については、『オペレーション』ドキュメントの「出力ファイルおよびワークファイルのサポート」を参照してください。

## BS2000/OSD バッチおよび TIAM でのワークファイル名

BS2000/OSD 環境で、アクセスメソッド AM=STD を指定してワークファイル番号を定義する場合は、*operand1* を使用して、このワークファイルに割り当てるファイル名またはリンク名を指定できます。

この場合、*operand1* は長さが 1～253 文字になり、次のいずれかを表します。

- BS2000/OSD でのリンク名 (1～8 文字)
- BS2000/OSD でのリンク名 (9～54 文字)
- **BS2000/OSD** での汎用ファイル名 (ワイルドカード)
- BS2000/OSD でのファイル名とリンク名
- **BS2000/OSD** での汎用ファイル名と汎用リンク名 (ワイルドカード)
- **\*DUMMY**

次の規則が適用されます。

1. ファイル名とリンク名は、位置指定パラメータとしてもキーワードパラメータとしても指定できます。対応するキーワードは、FILE= および LINK= です。位置指定パラメータとキーワードパラメータの混在は可能ですが、お勧めしません。
2. コンマなしの 1～8 文字長の文字列はリンク名と解釈されます。この表記は Natural の旧バージョンと互換性があります。例：

```
DEFINE WORK FILE 1 'W01'
```

キーワードパラメータを使用した対応する定義は次のとおりです。

```
DEFINE WORK FILE 1 'LINK=W01'
```

3. コンマなしの 9～54 文字長の文字列はファイル名と解釈されます。

例：

```
DEFINE WORK FILE 2 'NATURALvr.TEST.WORKFILE02'
```

*vr* は、Natural バージョンとリリース番号です。

キーワードパラメータを使用した対応する定義は次のとおりです。

```
DEFINE WORK FILE 2 'FILE=NATURALvr.TEST.WORKFILE02'
```

4. 次の入力、長さを考慮せずに解釈されるので、ルール 2 と 3 の例外です。

- キーワード入力：LINK=、FILE=
- \*DUMMY
- NULLFILE (\*DUMMY と同じ)
- \*
- \*,\*

例：DEFINE WORK FILE 7 'FILE=Y' は文字列に含まれる文字数が 9 文字未満ですが、有効なファイル割り当てであり、リンク名ではありません。

5. 一般的なファイル名の形式は次のとおりです。

```
Wnn.userid.tsn.date.time.number
```

上記の意味は次に示すとおりです。

<i>nn</i>	ワークファイル番号
<i>userid</i>	Natural ユーザー ID、8 文字
<i>tsn</i>	現在のタスクの BS2000/OSD TSN、4 桁
<i>date</i>	DDMMYYYY
<i>time</i>	HHIISS
<i>number</i>	番号、5 桁

6. 一般的なリンク名の形式は次のとおりです。

```
NWFnnnnn
```

*nnnnn* は 5 桁の数字で、ダイナミックリンク名が生成されるたびに 1 ずつ増加します。

7. ワークファイル番号へのファイル割り当てを変更すると、割り当てられていたワークファイルは CLOSE により暗黙的に閉じられます。

リンク名のみを指定する場合（例：W01）を除き、すべての場合においてキーワードパラメータを使用することをお勧めします。これは、追加レポートとの解釈の競合を防止し、9 文字未満のファイル名を使用する場合に必須です。



例：

```
DEFINE WORK FILE 3 'LINK=#W03'
DEFINE WORK FILE 3 'FILE=#W03'
```

<p>リンク名</p>	<p>例：</p> <pre>DEFINE WORK FILE 1 'LINKW01'</pre> <p>次の構文と同じ意味です。</p> <pre>DEFINE WORK FILE 1 'LINK=LINKW01'</pre> <p>LINK LINKW01 を含むファイルが、実行時に存在している必要があります。これは、Natural を開始する前に JCL を使用するか、または現在のアプリケーションからダイナミックに割り当てることにより、作成することができます。ダイナミックアロケーションには、ライブラリ SYSEXT にあるアプリケーションプログラミングインターフェイス USR2029N（ダイナミックファイルアロケーション）を使用できます。実行前に、リンクが別のファイル上で有効になっていた場合（例：W01）、プロファイルパラメータ FREE の値（指定可能な値は ON および OFF）に応じて、リンクは開放または保持されます。解放は、BS2000/OSD コマンドプロセッサに対する明示的な RELEASE コールによって実行されます。</p>
<p>ファイル名</p>	<p>例：</p> <pre>DEFINE WORK FILE 2 'NATURALvr.TEST.WORK02'</pre> <p>次の構文と同じ意味です。</p> <pre>DEFINE WORK FILE 2 'FILE=NATURALvr.TEST.WORK02'</pre> <p>vr は、Natural バージョンとリリース番号です。</p> <p>operand1 で指定したファイルは、FILE マクロコールを使用して設定され、また DEFINE WORK FILE ステートメントの実行前に対応するワークファイルに対して有効だったリンク名を継承します。</p>
<p>一般的なファイル名</p>	<p>例：</p> <pre>DEFINE WORK FILE 21 '*'</pre>

	<p>次の構文と同じ意味です。</p> <pre>DEFINE WORK FILE 21 'FILE=*'</pre> <p>ルール4に従って作成した名前を持つファイルは、FILEマクロコールを使用して設定され、また DEFINE WORK FILE ステートメントの実行前に対応するワークファイルに対して有効だったリンク名を継承します。</p> <pre>DEFINE WORK FILE 22 'FILE=*,LINK=WFLK22'</pre> <p>ルール4に従って作成した名前を持つファイルには、指定したリンク名が FILE マクロコールを使用して設定されます。</p>
<p>ファイル名とリンク名</p>	<p>例：</p> <pre>DEFINE WORK FILE 11 'NATURALvr.TEST.WORKF11,LNKW11'</pre> <p>次の構文と同じ意味です。</p> <pre>DEFINE WORK FILE 11 'FILE=NATURALvr.TEST.WORKF11,LINK=LNKW11'</pre> <p>次の構文も同じ意味です。</p> <pre>DEFINE WORK FILE 11 'FILE=NATURALvr.TEST.WORKF11,LNKW11'</pre> <p><i>vr</i> は、Natural バージョンとリリース番号です。</p> <p><i>operand1</i> で指定したファイルには、指定したリンク名が FILE マクロコールを使用して設定されます。そして、このファイルは対応するワークファイル番号に割り当てられます。</p>
<p>一般的なファイル名とリンク名</p>	<p>例：</p> <pre>DEFINE WORK FILE 27 '*,*'</pre> <p>次の構文と同じ意味です。</p> <pre>DEFINE WORK FILE 27 'FILE=*,LINK=*'</pre> <p>ルール4 およびルール5に従って作成したファイル名を持つファイルは、FILE マクロコールを使用して設定され、指定したワークファイル27に割り当てられます。</p> <p><b>注意:</b> ファイル名とリンク名を指定した場合、プロファイルパラメータ WORK のサブパラメータ FREE の値に関係なく、以前のリンク名は解放されません。</p>
<p>*DUMMY</p>	<p>ダミーのデータセットを示します。</p>

## CICS でのワークファイル名

アクセスメソッド AM=CICS を指定してワークファイル番号を定義する場合は、ワークファイルのプロファイルパラメータ WORK のサブパラメータ TYPE に合わせて、*operand1* を一時データまたは一時ストレージキューの名前（1~8文字）にすることができます。TYPE=TDでは、*operand1* の最初の4文字だけが保持されるので、事前に一時データの送り先として CICS を指定しておく必要があります。

ワークファイルの詳細については、「CICS 環境での *Natural* の出力ファイルおよびワークファイル」（『TP モニタインターフェイス』ドキュメント）も参照してください。

## Com-plete/SMARTS でのワークファイル名

Com-plete 環境で AM=SMARTS を指定すると、PFS ファイルを利用できます。まだワークファイル名が *Natural* に定義されていない場合でも、任意のワークファイル名を割り当てることができます。次に例を示します。

```
DEFINE WORK (14) '/nat/path/workfile'  
DEFINE WORK (14) 'workfile'
```

ファイルが SMARTS ポータブルファイルシステムまたはネイティブファイルシステムのどちらにあるかは、SMARTS の MOUNT\_FS パラメータによって異なります。パスの最初の要素 (*/nat/*) により、ターゲットファイルシステムが決定されます。

文字列の先頭がスラッシュ (/) でない場合は、ファイルのパスが環境変数 \$NAT\_WORK\_ROOT から取得されます。

指定したパス名が存在する必要があります。DEFINE WORK FILE ステートメントを実行すると、ファイルがダイナミックに割り当てられます。指定したメンバが存在しない場合は、指定した名前の新しいメンバが作成されます。



# 52 DELETE

---

▪ 機能 .....	316
▪ 制限事項 .....	316
▪ 構文説明 .....	316
▪ データベース固有の考慮事項 .....	317
▪ 例 .....	317

```
DELETE [RECORD] [IN] [STATEMENT] [(r)]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION DATA](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

DELETE ステートメントは、データベースからレコードを削除するために使用します。

### ホールド状態

DELETE ステートメントを使用すると、対応する [FIND](#) や [READ](#) ステートメントで選択した各レコードはホールド状態に置かれます。

レコードのホールドロジックの詳細については、『[プログラミングガイド](#)』の「[データベース更新- トランザクション処理](#)」セクションを参照してください。

## 制限事項

DELETE ステートメントは、[FIND](#)、[READ](#)、または [GET](#) ステートメントと同じ行に指定することはできません。

## 構文説明

<b>(r)</b>	ステートメント参照： 表記 (r) を指定して、削除するレコードの選択と読み取りに使用されたステートメントを参照します。  (r) を指定しなかった場合、DELETE ステートメントは、データベースレコードが選択されて読み込まれた最も内側でアクティブになっている処理ループを参照します。
------------	--

## データベース固有の考慮事項

DL/I データベース	DELETE ステートメントは、DL/I データベースからセグメントを削除するために使用します。そのセグメントに従属するすべてのものが削除されます。  GSAM 制限により、UPDATE ステートメントは GSAM データベースには使用できません。
VSAM データベース	DELETE ステートメントは、VSAM エントリ順データセット (ESDS) に対しては使用できません。
SQL データベース	DELETE ステートメントは、データベーステーブルから 1 つの行を削除するために使用します。これは、SQL ステートメントの DELETE WHERE CURRENT OF CURSOR-NAME に対応するものです。つまり、最後に読み取られた行のみを削除できます。  ほとんどの SQL データベースでは、FIND SORTED BY または READ LOGICAL ステートメントで読み取られた行は削除できません。

## 例

- 例 1
- 例 2

### 例 1

この例では、ALDEN を含むレコードがすべて削除されます。

```

** Example 'DELEX1': DELETE
**
**
CAUTION: Executing this example will modify the database records!
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
END-DEFINE
*
FIND EMPLOY-VIEW WITH NAME = 'ALDEN'
/*
DELETE
END TRANSACTION
/*
AT END OF DATA
  WRITE NOTITLE *NUMBER 'RECORDS DELETED'
END-ENDDATA

```

## DELETE

---

```
END-FIND  
END
```

### 例 2

VEHICLES ファイルで ALDEN という名前の人物が見つからない場合は、ALDEN の EMPLOYEE レコードが削除されます。

```
** Example 'DELEX2': DELETE  
**  
**  
CAUTION: Executing this example will modify the database records!  
*****  
DEFINE DATA LOCAL  
1 EMPLOY-VIEW VIEW OF EMPLOYEES  
  2 PERSONNEL-ID  
  2 NAME  
1 VEHIC-VIEW VIEW OF VEHICLES  
  2 PERSONNEL-ID  
END-DEFINE  
*  
EMPL. FIND EMPLOY-VIEW WITH NAME = 'ALDEN'  
  /*  
  VEH. FIND VEHIC-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (EMPL.)  
  IF NO RECORDS FOUND  
    /*  
    DELETE (EMPL.)  
    /*  
  END TRANSACTION  
  END-NOREC  
END-FIND  
  /*  
END-FIND  
END
```



# 53      DISPLAY

---

▪ 機能 .....	320
▪ 構文説明 .....	320
▪ DISPLAY ステートメントに適用されるデフォルト .....	332
▪ 例 .....	333

```
DISPLAY [(rep)] [options] [/...][output-format] output-element] ...
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT END OF PAGE](#) | [AT TOP OF PAGE](#) | [CLOSE PRINTER](#) | [DEFINE PRINTER EJECT](#) | [FORMAT](#) | [NEWPAGE](#) | [PRINT](#) | [SKIP](#) | [SUSPEND IDENTICAL SUPPRESS](#) | [WRITE](#) | [WRITE TITLE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

DISPLAY ステートメントでは、レポートに出力するフィールドを列形式で指定します。列は各フィールドごとに作成され、フィールドヘッダーが列の上に表示されます。



**注意:** WRITE ステートメントと PRINT ステートメントを使用すると、フリーフォーマット（列なし）の出力を生成することができます。

『プログラミングガイド』の次のトピックも参照してください。

- データ出力制御
- DISPLAY および WRITE ステートメント
- マルチプルバリューフィールドとピリオディックグループのインデックス表記
- 列ヘッダー
- 出力ページのレイアウト

## 構文説明

<i>(rep)</i>	<p>レポート指定：</p> <p>表記 (<i>rep</i>) を使用すると、DISPLAY ステートメントを適用するレポートの ID を指定できます。</p> <p>レポート ID として、範囲 0~31 の値、または <a href="#">DEFINE PRINTER</a> ステートメントを使用して割り当てた論理名を指定できます。</p> <p>(<i>rep</i>) を指定しない場合、ステートメントは最初のレポート（レポート 0）に適用されます。</p> <p>プリンタファイルが PC として Natural に定義されている場合、レポートは PC にダウンロードされます。「<a href="#">例 8</a>」を参照してください。</p>
--------------	---

	Natural で作成される出力レポートの形式を制御する方法については、『プログラミングガイド』の「データ出力制御」を参照してください。
<i>options</i>	表示オプション： 詳細については、下記の「 <a href="#">表示オプション</a> 」を参照してください。
<i>output-format</i>	出力フォーマット定義： 詳細については、下記の「 <a href="#">出力フォーマット定義</a> 」を参照して下さい。
/	行送り - スラッシュ表記： 文字列内でスラッシュ (/) を指定すると、表示中の文字列が改行されます。 出力項目の間にスラッシュを指定すると、スラッシュ (/) で指定された出力項目は、同じ列内で縦方向に配置されます。この列のヘッダーでは、縦方向に表示されている項目のヘッダーが、列の上側に縦方向で配置されます。 『プログラミングガイド』の次のトピックも参照してください。 ■ 行送り - スラッシュ表記 ■ 例1 - DISPLAY ステートメントでの改行 ■ 列ヘッダーの省略 - スラッシュ表記
<i>output-element</i>	出力項目： 詳細については、下記の「 <a href="#">出力項目</a> 」を参照してください。

## 表示オプション

```
[NOTITLE] [NOHDR] [ [AND] [GIVE] [SYSTEM] FUNCTIONS ] [(statement-parameters)]
```

構文要素の説明：

<b>NOTITLE</b>	<p>デフォルトページタイトルの省略：</p> <p>デフォルトでは、DISPLAY ステートメントで出力される各ページに、Natural によって 1 行のタイトル行が作成されます。このタイトルには、ページ番号と日時が含まれます。時刻は、プログラム実行開始時 (TP モード) またはジョブ開始時 (バッチモード) に設定されます。このデフォルトタイトル行は、<a href="#">WRITE TITLE</a> ステートメントで書き換えることができます。また、DISPLAY ステートメントにキーワード NOTITLE を指定して、タイトル行を省略することもできます。</p> <p>例：</p>
----------------	---

	<ul style="list-style-type: none"> <li>■ デフォルトタイトルが生成されます。</li> </ul> <pre style="background-color: #f0f0f0; padding: 5px;">DISPLAY NAME</pre> <ul style="list-style-type: none"> <li>■ ユーザータイトルが生成されます。</li> </ul> <pre style="background-color: #f0f0f0; padding: 5px;">DISPLAY NAME WRITE TITLE 'user-title'</pre> <ul style="list-style-type: none"> <li>■ タイトルは生成されません。</li> </ul> <pre style="background-color: #f0f0f0; padding: 5px;">DISPLAY NOTITLE NAME</pre> <p><b>注意:</b> NOTITLE オプションを使用すると、同じオブジェクト内でデータを同じレポートに書き込むすべての DISPLAY、PRINT、および WRITE ステートメントに適用されます。</p>
<p><b>NOHDR</b></p>	<p>列ヘッダー：</p> <p>列ヘッダーは、DISPLAY ステートメントで指定した各フィールド（列）ごとに、次の規則に従って生成されます。</p> <ul style="list-style-type: none"> <li>■ DISPLAY ステートメントでは、フィールド名の前に表示されるヘッダー文字列を明示的に指定できます。次に例を示します。</li> </ul> <pre style="background-color: #f0f0f0; padding: 5px;">DISPLAY 'EMPLOYEE' NAME 'SALARY' SALARY</pre> <ul style="list-style-type: none"> <li>■ フィールドのヘッダーを明示的に指定しない場合は、<b>DEFINE DATA</b> ステートメントで定義した<b>ヘッダー</b>が使用されます。</li> <li>■ DEFINE DATA ステートメントでデータベースフィールドのヘッダーが定義されていない場合は、DDMに定義されているヘッダーがデフォルトとして使用されます。</li> <li>■ デフォルトのヘッダーがDDMに定義されていない場合は、フィールド名がヘッダーとして使用されます。</li> <li>■ DEFINE DATA ステートメントでユーザー定義変数のヘッダーが定義されていない場合は、変数名がヘッダーとして使用されます。<a href="#">ヘッダー定義</a>については、DEFINE DATA ステートメントも参照してください。</li> </ul> <pre style="background-color: #f0f0f0; padding: 5px;">DISPLAY NAME SALARY #NEW-SALARY</pre> <ul style="list-style-type: none"> <li>■ <b>Natural</b> では、常に列ヘッダーに下線が付けられ、表示するデータと下線の間に 1 行の空行が生成されます。</li> <li>■ プログラム内に複数の DISPLAY ステートメントがある場合は、最初の DISPLAY ステートメントにより、使用する列ヘッダーが決定されます。これはコンパイル時に評価されます。</li> </ul> <p>列ヘッダーの省略：</p>

	<p>1つのフィールドの列ヘッダーを省略するには</p> <ul style="list-style-type: none"> <li>■ フィールド名の前に、次の文字列（アポストロフィ-スラッシュ-アポストロフィ）を指定します。</li> </ul> <pre>'/'</pre> <p>例えば、次のようになります。</p> <pre>DISPLAY '/' NAME 'SALARY' SALARY</pre> <p>すべての列ヘッダーを省略するには</p> <ul style="list-style-type: none"> <li>■ キーワード NOHDR を指定します。</li> </ul> <pre>DISPLAY NOHDR NAME SALARY</pre> <p><b>注意:</b></p> <ol style="list-style-type: none"> <li>1. NOHDRは最初の DISPLAY ステートメントにのみ有効です。これは、後続の DISPLAY ステートメントでは列ヘッダーを作成できないためです。</li> <li>2. NOTITLE と NOHDR の両方を使用する場合は、DISPLAY NOTITLE NOHDR NAME SALARY の順で指定する必要があります。</li> </ol>
<b>GIVE SYSTEM FUNCTIONS</b>	<p>システム関数の使用：</p> <p>GIVE SYSTEM FUNCTIONS 節を使用すると、<b>Natural</b> システム関数 AVER、COUNT、MAX、MIN、NAVER、NCOUNT、NMIN、SUM、TOTAL が利用できるようになります。これらのシステム関数は、GIVE SYSTEM FUNCTIONS 節を含む DISPLAY ステートメントの実行時に評価されます。</p> <p>これらのシステム関数は、AT END OF PAGE 条件で実行されるステートメントで参照できます。</p> <p><b>注意:</b></p> <ol style="list-style-type: none"> <li>1. レポートごとに、1つの DISPLAY ステートメントにのみ GIVE SYSTEM FUNCTIONS 節を含めることができます。DISPLAY ステートメントからシステム関数が評価されるたびに、システム関数はページ単位で評価されます。つまり、新しいページが始まると、TOTAL 以外の関数はすべてゼロにリセットされます。</li> <li>2. システム関数をサブルーチン内の DISPLAY ステートメントで使用する場合は、END OF PAGE 処理を同じルーチン内で開始する必要があります。</li> </ol> <p>「例 2 - DISPLAY ステートメントでの GIVE SYSTEM FUNCTIONS 節の使用例」も参照してください。</p>
<i>statement-parameters</i>	ステートメントレベルでのパラメータ定義：

	<p>1つまたは複数のパラメータをカッコで囲んで、要素（フィールド）レベルで、つまり DISPLAY の直後に指定できます。</p> <p>この方法で指定した各パラメータによって、以前に GLOBALS コマンド、<b>SET GLOBALS</b> ステートメント（レポートモードのみ）、または <b>FORMAT</b> ステートメントで指定した対応するパラメータは上書されます。</p> <p>複数のパラメータを指定する場合は、個々のパラメータを1つ以上の空白で区切る必要があります。各パラメータ指定を2行のステートメント行に分割することはできません。</p> <p><b>注意:</b> ここで適用されるパラメータ設定は、変数フィールドにのみ関連し、テキスト定数には影響しません。テキスト定数にフィールド属性を設定する場合は、この要素に属性を明示的に設定する必要があります（「<b>要素（フィールド）レベルでのパラメータ定義</b>」を参照）。</p> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <b>パラメータのリスト</b></li> <li>■ <b>ステートメントおよび要素（フィールド）レベルでのパラメータ使用例</b></li> <li>■ <b>例7-ステートメント／要素レベルでパラメータを使用した DISPLAY ステートメント</b></li> </ul>
--	---

## パラメータのリスト

DISPLAY ステートメントで指定可能なパラメータ		指定 (S = ステートメントレベル、E = 要素レベル)
AD	属性定義	SE
AL	出力の英数字長	SE
BX	ボックス定義	SE
CD	カラー定義	SE
CV	制御変数	SE
DF	日付フォーマット	SE
DL	出力の表示長	SE
DY	ダイナミック属性	SE
EM	編集マスク	SE
ES	空行の省略	S
FC	充填文字	SE
FL	浮動小数点仮数長	SE
GC	グループヘッダーの充填（フィラー）文字	SE
HC	ヘッダーの中央揃え	SE
HW	ヘッダーの幅	SE

DISPLAY ステートメントで指定可能なパラメータ		指定 (S = ステートメントレベル、E = 要素レベル)
IC	挿入文字	SE
IS	重複抑制	SE
LC	先頭文字	SE
LS	行サイズ	S
MC	マルチプルバリューフィールド数	S
MP	レポートの最大ページ数	S
NL	出力の数値長	SE
PC	ピリオディックグループ数	S
PM	出力モード	SE
PS	ページサイズ	S
SF	フィールド間の空白	SE
SG	符号の位置	SE
TC	末尾文字	SE
UC	下線付き文字	SE
ZP	ゼロ出力	SE

各パラメータの詳細については、『パラメータリファレンス』（セッションパラメータ）を参照してください。

『プログラミングガイド』の次のトピックも参照してください。

- 列ヘッダーの中央揃え - HC パラメータ
- 列ヘッダーの幅 - HW パラメータ
- ヘッダーの充填文字 - FC および GC パラメータ
- タイトルおよびヘッダーの下線付き文字 - UC パラメータ

## ステートメントおよび要素（フィールド）レベルでのパラメータ使用例

```

DEFINE DATA LOCAL
1 VARI (A4)      INIT <'1234'>          /*      Output
END-DEFINE      /*      Produced
*              /*      -----
DISPLAY NOHDR   'Text'                  '='  VARI          /*      Text 1234
DISPLAY NOHDR (PM=I) 'Text'              '='  VARI          /*      Text 4321
DISPLAY NOHDR   'Text' (PM=I)           '='  VARI (PM=I) /*      txeT 4321
DISPLAY NOHDR   'Text' (PM=I)           '='  VARI          /*      txeT 1234
END

```

## 出力フォーマット定義

$\left[ \begin{array}{l} nX \\ nT \\ x/y \\ T*field-name \\ P*field-name \end{array} \right]$	$\left[ \begin{array}{l} 'text' [(attributes)] \\ 'c' (n) [(attributes)] \end{array} \right] \dots$
$\left[ \begin{array}{l} \text{VERTICALLY} \\ \text{HORIZONTALLY} \end{array} \right]$	$\left[ \text{AS} \left\{ \begin{array}{l} 'text' [(attributes)] [\text{CAPTIONED}] \\ [\text{CAPTIONED}] \end{array} \right\} \right] [/\dots]$

### フィールドの位置指定表記

<b>nX</b>	<p>列の間隔：</p> <p>この表記により、列の間に <math>n</math> 個のスペースが挿入されます。<math>n</math> をゼロにすることはできません。</p> <p>例：</p> <pre>DISPLAY NAME 5X SALARY</pre> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <a href="#">例 1 - nX および nT 表記を使用した DISPLAY ステートメント</a> (後述)</li> <li>■ <a href="#">列の間隔 - SF パラメータと nX 表記</a> (『プログラミングガイド』)</li> </ul>
<b>nT</b>	<p>タブ設定：</p> <p><math>nT</math> 表記により、位置 <math>n</math> を表示するように位置指定 (タブ設定) されます。後方への位置指定はできません。</p> <p>次の例では、NAME が 25 桁目から、SALARY が 50 桁目から表示されます。</p> <pre>DISPLAY 25T NAME 50T SALARY</pre> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <a href="#">例 1 - nX および nT 表記を使用した DISPLAY ステートメント</a> (後述)</li> <li>■ <a href="#">タブ設定 - nT 表記</a> (『プログラミングガイド』)</li> </ul>



$x/y$	<p><b><math>x/y</math> 位置指定：</b></p> <p><math>x/y</math> 表記により、次の要素は最後のステートメント出力の <math>x</math> 行下、列 <math>y</math> の先頭に配置されます。<math>y</math> をゼロにすることはできません。後方への位置指定はできません。</p>
<b>T*</b> <i>field-name</i>	<p><b>フィールドの相対位置指定：</b></p> <p>T* 表記により、前の DISPLAY ステートメントで使用されたフィールドの特定の表示桁を位置指定できます。後方への位置指定はできません。</p>
<b>P*</b> <i>field-name</i>	<p><b>フィールドおよび行の相対位置指定：</b></p> <p>P* 表記により、前の DISPLAY ステートメントで使用されたフィールドの特定の表示桁を位置指定できます。縦 (VERT) 方向の表示モードと組み合わせて使用します。後方への位置指定はできません。</p> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <b>例3 - P* 表記を使用した DISPLAY ステートメント</b> (後述)</li> <li>■ <b>タブ表記 P*<i>field</i></b> (『プログラミングガイド』)</li> </ul>

### 列ヘッダー割り当ての上書き

<b>'text'</b>	<p><b>テキスト割り当て：</b></p>
<b>'/'</b>	<p>フィールドの直前に指定すると、一重引用符で囲まれた文字列によって列ヘッダーが上書きされます。</p> <p>フィールド名の前にスラッシュ文字 '/' を指定すると、列ヘッダーは省略されます。</p> <pre>DISPLAY 'EMPLOYEE' NAME 'MARITAL/STATUS' MAR-STAT</pre> <p>複数の 'text' 要素をフィールド名の前に指定すると、最後の 'text' 要素が列ヘッダーとして使用され、残りの要素は列内でフィールド値の前に配置されます。</p> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <b>独自の列ヘッダーの定義</b> (『プログラミングガイド』)</li> <li>■ <b>「テキスト表記」、<b>「ステートメントで使用するテキストの定義</b>」</b> (『プログラミングガイド』)</li> <li>■ <b>例4 - 'text'、'c(n)'、および属性表記を使用した DISPLAY ステートメント</b> (後述)</li> </ul>
<b>'c'(n)</b>	<p><b>文字の繰り返し：</b></p>

フィールド値の直前に、一重引用符で囲まれた文字列が  $n$  回表示されます。次に例を示します。

```
DISPLAY '*' (5) '=' NAME
```

結果


```
***** SMITH
```

以下の項目も参照してください。

- テキスト表記、フィールド値の前に  $n$  回表示される文字の定義（『プログラミングガイド』）
- [例 4 - 'text'、'c\(n\)'、および属性表記を使用した DISPLAY ステートメント](#)（後述）


### 出力属性

`attributes` は、テキスト表示に使用される出力属性を示します。可能な属性：



指定可能なセッションパラメータ値については、『パラメータリファレンス』ドキュメントの該当するセクションを参照してください。

- 「AD - 属性定義」の「フィールド表現」
- CD - カラー定義
- BX - 外枠の定義
- PM - 出力モード

 **注意:** コンパイラは、実際には1つの出力フィールドに複数の属性値を受け入れます。例えば、「AD=BDI」と指定できます。ただし、この場合は最後の値のみが適用されます。示した例では、値 I のみが有効になり、出力フィールドは強調表示されます。

### VERTICAL（縦）／HORIZONTAL（横）表示

VERT 節は、複数のフィールド値を同一の列内で上下に表示するために使用します。VERT モードでは、キーワード VERT または HORIZ を指定すると新しい列が開始されます。

VERT モードの列ヘッダーは、次に示すように AS 節で指定したエントリで制御されます。

<b>VERTICALLY</b>	<p>縦方向の列。AS 節を省略すると、列ヘッダーは出力されません。</p> <pre>DISPLAY VERT NAME SALARY</pre> <p>表示例については、『プログラミングガイド』の「AS 節のない DISPLAY VERT」を参照してください。</p>
<b>AS 'text'</b>	<p>縦方向の列。AS 'text' を指定すると、一重引用符内の文字列が列ヘッダーとして使用されます。</p> <p>表示例については、『プログラミングガイド』の「DISPLAY VERT AS 'text'」を参照してください。</p> <p>'text' の文字列中にスラッシュ文字 / がある場合、列ヘッダーは改行されて複数行になります。</p> <pre>DISPLAY VERT AS 'LAST/NAME' NAME</pre>
<b>AS 'text' CAPTIONED</b>	<p>縦方向の列。AS 'text' CAPTIONED を指定すると、'text' が列ヘッダーとして使用され、各明細行のフィールド値の直前に標準ヘッダーまたはフィールド名が挿入されます。</p> <pre>DISPLAY VERT AS 'PERSONS/SELECTED' CAPTIONED NAME FIRST-NAME</pre> <p>表示例については、『プログラミングガイド』の「DISPLAY VERT AS 'text' CAPTIONED」を参照してください。</p>
<b>AS CAPTIONED</b>	<p>縦方向の列。AS CAPTIONED を指定すると、フィールドの標準ヘッダー（ヘッダー文字列あるいはフィールド名）が列ヘッダーとして使用されます。</p> <pre>DISPLAY VERT AS CAPTIONED NAME FIRST-NAME</pre>
<b>HORIZONTALLY</b>	<p>横方向の列。これはデフォルトの表示モードです。</p>

キーワードを使用して縦方向の表示と横方向の表示を混在させることができます。

ある出力項目の縦方向の表示を一時的に停止する必要がある場合は、その要素の前に "-"（ハイフン）を指定します。次に例を示します。

```
DISPLAY VERT NAME - FIRST-NAME SALARY
```

この例では、FIRST-NAME は NAME の横に横方向で表示され、SALARY は NAME の下に縦方向で表示されます。

標準の表示モードは、横方向です。表示されるフィールドごとに 1 つの列が生成されます。

列ヘッダーは、Natural によって次の優先順位で取得および使用されます。

1. DISPLAY ステートメントで指定したヘッダーの 'text'。
2. データベースフィールドについては DDM に定義されたデフォルトのヘッダー、またはユーザー定義変数については変数名。
3. データベースフィールドに対してヘッダー文字列が定義されていない場合は、DDM に定義されたフィールド名。

グループ名については、グループ全体に 1 つのグループヘッダーがつけられます。グループを指定する場合、グループ全体に対するヘッダーだけをユーザーが変更できます。

列ヘッダーの最大行数は 15 行です。

DISPLAY ステートメントからの出力は、行の幅（ラインサイズ）を超えないようにする必要があります。行の幅を超えると、エラーメッセージが発行されます。

縦方向の表示と横方向の表示の詳細な使用方法については、次の例を参照してください。

- [例 5 - 横方向の表示を使用した DISPLAY ステートメント](#)
- [例 6 - 縦および横方向の表示を使用した DISPLAY ステートメント](#)
- `DISPLAY VERT AS CAPTIONED` および `HORIZ`（『プログラミングガイド』）

## 出力項目

```

[ { 'text' [(attributes)] } ...
  { 'c'(n) [(attributes)] } ...
  nX
  nT
  x/y
] ['='] {operand1 [(parameters)]}
    
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<code>operand1</code>	S A G N A N P I F B D T L	G O	可	不可

## 構文要素の説明

<i>nX</i>	列の間隔： これは、「出力フォーマット定義」の場合と同じです（ <a href="#">上記参照</a> ）。
<i>nT</i>	タブ設定： これは、「出力フォーマット定義」の場合と同じです（ <a href="#">上記参照</a> ）。
<i>x/y</i>	<i>x/y</i> 位置指定： これは、「出力フォーマット定義」の場合と同じです（ <a href="#">上記参照</a> ）。
'text'	テキスト割り当て： これは、「出力フォーマット定義」の場合と同じです（ <a href="#">上記参照</a> ）。
'c'(n)	文字の繰り返し： これは、「出力フォーマット定義」の場合と同じです（ <a href="#">上記参照</a> ）。
'text' '='	'text' '=' をフィールドの直前に置くと、'text' がフィールド値の直前に出力されます。 これは、'c' (n) '=' と同じように適用できます。
'c' (n) '='	DISPLAY '*****' '=' NAME
<i>attributes</i>	出力属性： これは、「出力フォーマット定義」の場合と同じです（ <a href="#">上記参照</a> ）。
<i>operand1</i>	表示フィールド。 <b>注意:</b> DL/I データベースの場合：DL/I AIX フィールドは、パラメータ PROCSEQ で指定した AIX とともに PCB を使用している場合にのみ表示できます。それ以外の場合は、実行時に Natural によってエラーメッセージが返されます。
<i>parameters</i>	要素（フィールド）レベルでのパラメータ定義： 1 つまたは複数のパラメータをカッコで囲んで、要素（フィールド）レベルで、つまり <i>operand1</i> の直後に指定できます。この方法で指定した各パラメータは、以前に <a href="#">ステートメントレベル</a> 、GLOBALs コマンド、 <a href="#">SET GLOBALS</a> ステートメント（レポートモードでのみ）、または <a href="#">FORMAT</a> ステートメントで指定した対応するパラメータを上書きします。 複数のパラメータを指定する場合は、各エントリ間に 1 つ以上の空白を配置する必要があります。エントリを 2 行のステートメント行に分割することはできません。 以下の項目も参照してください。 ■ <a href="#">パラメータのリスト</a> ■ <a href="#">ステートメントおよび要素（フィールド）レベルでのパラメータ使用例</a>

## DISPLAY ステートメントに適用されるデフォルト

DISPLAY ステートメントには、次のデフォルトが適用されます。

- レポートの幅  
レポートの幅は、Natural をインストールしたときの設定値がデフォルトになります。このデフォルト値は通常、バッチモードでは 132 であり、TP モードでは端末の 1 行の長さです。この値は、セッションパラメータ LS で上書きできます。TP モードでは、行の長さ (LS) とページの大きさ (PS) のパラメータは、使用中の端末タイプの物理特性に基づいて Natural で設定されます。
- 端末画面の出力  
DISPLAY ステートメントの出力を端末画面に表示する場合、出力は物理的に 2 桁目から始まります。これは、1 桁目が 3270 タイプ端末の属性桁として予約されているためです。
- 用紙への印刷  
DISPLAY ステートメントの出力を用紙に印刷する場合は、1 桁目から印刷されます。
- フィールド間の空白  
出力項目間のデフォルトのフィールド間の空白は 1 桁です。したがって、列の間には最小 1 桁の空白があげられます (端末属性用に予約済)。この値は、セッションパラメータ SF で上書きできます。
- フィールド出力  
フィールド長、またはフィールドヘッダーのうちいずれか長い方によって、レポートの列の幅が決まります (HW パラメータが使用されない場合)。
  - フィールドの方がヘッダーよりも長い場合、パラメータ HC=L または HC=R でヘッダーを左寄せまたは右寄せにする指示がない限り、ヘッダーは列の上でセンタリングされます。
  - また、ヘッダーの方がフィールドより長い場合、フィールドはヘッダーの下に左詰めで出力されます。
  - フィールド内の値は、英数字フィールドでは左詰めにになり、数字フィールドでは右詰めにになります。
  - 数値フィールドは、AD=L を指定すると左詰めで表示されます。
  - 英数字フィールドは、AD=R を指定すると右詰めで表示できます。
  - 縦方向の表示では、フィールドの中で最も長いデータ値またはヘッダーによって列の幅が決まります (HW パラメータを使用しない場合)。
- 符号  
数字フィールドの印刷時には、符号桁として 1 桁余分に左側に確保されます。Natural セッションパラメータ SG を使用すると、符号桁を省略できます。

## ■ ページオーバーフロー

DISPLAY ステートメントの実行前に、ページオーバーフローがチェックされます。DISPLAY ステートメントの実行中に、新しいページタイトルまたはトレイラ情報が生成されることはありません。

## 例

- 例 1 - nX および nT 表記を使用した DISPLAY ステートメント
- 例 2 - DISPLAY ステートメントでの GIVE SYSTEM FUNCTIONS 節の使用例
- 例 3 - P\* 表記を使用した DISPLAY ステートメント
- 例 4 - 'text'、'c(n)'、および属性表記を使用した DISPLAY ステートメント
- 例 5 - 横方向の表示を使用した DISPLAY ステートメント
- 例 6 - 縦および横方向の表示を使用した DISPLAY ステートメント
- 例 7 - ステートメント／要素レベルでパラメータを使用した DISPLAY ステートメント
- 例 8 - 出力ファイルを PC として Natural に定義する場合のレポート指定

### 例 1 - nX および nT 表記を使用した DISPLAY ステートメント

```
** Example 'DISEX1': DISPLAY (with nX, nT notation)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 JOB-TITLE
END-DEFINE
*
LIMIT 4
READ EMPL-VIEW BY NAME
  DISPLAY NOTITLE 5X NAME 50T JOB-TITLE
END-READ
*
END
```

プログラム **DISEX1** の出力：

NAME	CURRENT POSITION
-----	-----
ABELLAN	MAQUINISTA
ACHIESON	DATA BASE ADMINISTRATOR
ADAM	CHEF DE SERVICE
ADKINSON	PROGRAMMER

**例 2 - DISPLAY ステートメントでの GIVE SYSTEM FUNCTIONS 節の使用例**

```

** Example 'DISEX2': DISPLAY (with GIVE SYSTEM FUNCTIONS)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
  2 SALARY (1)
  2 CURR-CODE (1)
END-DEFINE
*
LIMIT 15
FORMAT PS=15
*
READ EMPLOY-VIEW
  DISPLAY GIVE SYSTEM FUNCTIONS
    PERSONNEL-ID NAME FIRST-NAME SALARY (1) CURR-CODE (1)
  AT END OF PAGE
    WRITE / 'SALARY STATISTICS:'
      / 7X 'MAXIMUM:' MAX(SALARY(1)) CURR-CODE (1)
      / 7X 'MINIMUM:' MIN(SALARY(1)) CURR-CODE (1)
      / 7X 'AVERAGE:' AVER(SALARY(1)) CURR-CODE (1)
  END-ENDPAGE
END-READ
*
END

```

プログラム **DISEX2** の出力：

PERSONNEL ID	NAME	FIRST-NAME	ANNUAL SALARY	CURRENCY CODE
50005500	BLOND	ALEXANDRE	172000	FRA
50005300	MAIZIERE	ELISABETH	166900	FRA
50004900	CAOUDAL	ALBERT	167350	FRA
50004600	VERDIE	BERNARD	170100	FRA
50004200	VAUZELLE	BERNARD	159790	FRA
50004100	CHAPUIS	ROBERT	169900	FRA
50003800	JOUSSELIN	DANIEL	171990	FRA
50006900	BAILLET	PATRICK	188000	FRA
50007600	MARX	JEAN-MARIE	365700	FRA

SALARY STATISTICS:				
MAXIMUM:	365700	FRA		
MINIMUM:	159790	FRA		
AVERAGE:	192414	FRA		



### 例 3 - P\* 表記を使用した DISPLAY ステートメント

```

** Example 'DISEX3': DISPLAY (with P* notation)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 SALARY (1)
  2 BIRTH
  2 CITY
END-DEFINE
*
LIMIT 2
READ EMPL-VIEW BY CITY FROM 'N'
  DISPLAY NOTITLE NAME CITY
    VERT AS 'BIRTH/SALARY' BIRTH (EM=YYYY-MM-DD) SALARY (1)
  SKIP 1
  AT BREAK OF CITY
    DISPLAY P*SALARY (1) AVER(SALARY (1))
    SKIP 1
  END-BREAK
END-READ
END

```

プログラム **DISEX3** の出力：

NAME	CITY	BIRTH	SALARY
-----	-----	-----	-----
WILCOX	NASHVILLE	1970-01-01	38000
MORRISON	NASHVILLE	1949-07-10	36000
			37000

### 例 4 - 'text'、'c(n)'、および属性表記を使用した DISPLAY ステートメント

```

** Example 'DISEX4': DISPLAY (with 'c(n)' notation and attribute)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 DEPT
  2 LEAVE-DUE
  2 NAME
END-DEFINE
*

```

## DISPLAY

```
LIMIT 4
READ EMPL-VIEW BY DEPT FROM 'T'
  IF LEAVE-DUE GT 40
    DISPLAY NOTITLE
      'EMPLOYEE' NAME           /* OVERRIDE STANDARD HEADER
      'LEAVE ACCUMULATED' LEAVE-DUE /* OVERRIDE STANDARD HEADER
      '**'(10)(I)              /* DISPLAY 10 '** INTENSIFIED

  ELSE
    DISPLAY NAME LEAVE-DUE
  END-IF
END-READ
*
END
```

プログラム **DISEX4** の出力：

EMPLOYEE	LEAVE ACCUMULATED
LAVENDA	33
BOYER	33
CORREARD	45 *****
BOUVIER	19

### 例 5 - 横方向の表示を使用した DISPLAY ステートメント

```
** Example 'DISEX5': DISPLAY (horizontal display)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 JOB-TITLE
  2 SALARY (1:2)
  2 CURR-CODE (1:2)
END-DEFINE
*
LIMIT 4
READ EMPL-VIEW BY NAME
  DISPLAY NOTITLE NAME JOB-TITLE SALARY (1:2) CURR-CODE (1:2)
  SKIP 1
END-READ
*
END
```

プログラム **DISEX5** の出力 :

NAME	CURRENT POSITION	ANNUAL SALARY	CURRENCY CODE
ABELLAN	MAQUINISTA	1450000 1392000	PTA PTA
ACHIESON	DATA BASE ADMINISTRATOR	11300 10500	UKL UKL
ADAM	CHEF DE SERVICE	159980 0	FRA 0
ADKINSON	PROGRAMMER	34500 31700	USD USD

### 例 6 - 縦および横方向の表示を使用した DISPLAY ステートメント

```

** Example 'DISEX6': DISPLAY (vertical and horizontal display)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CITY
  2 JOB-TITLE
  2 SALARY (1:2)
  2 CURR-CODE (1:2)
END-DEFINE
*
LIMIT 1
READ EMPL-VIEW BY NAME
  DISPLAY NOTITLE VERT AS CAPTIONED
    NAME CITY 'POSITION' JOB-TITLE
    HORIZ 'SALARY' SALARY (1:2) 'CURRENCY' CURR-CODE (1:2)
  /*
  SKIP 1
END-READ
END

```

プログラム **DISEX6** の出力：

NAME CITY POSITION	SALARY	CURRENCY
ABELLAN	1450000	PTA
MADRID	1392000	PTA
MAQUINISTA		

**例7-ステートメント／要素レベルでパラメータを使用した DISPLAY ステートメント**

```

** Example 'DISEX7': DISPLAY (with parameters for statement/element)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 PERSONNEL-ID
  2 TELEPHONE
    3 AREA-CODE
    3 PHONE
END-DEFINE
*
LIMIT 3
READ EMPL-VIEW BY NAME
  DISPLAY NOTITLE (AL=16 GC=+ NL=8 SF=3 UC=)
    PERSONNEL-ID NAME TELEPHONE (LC=< TC=>)
END-READ
END
    
```

プログラム **DISEX7** の出力：

PERSONNEL ID	NAME	AREA CODE	TELEPHONE
60008339	ABELLAN	<1 >	<4356726 >
30000231	ACHIESON	<0332 >	<523341 >
50005800	ADAM	<1033 >	<44864858 >

## 例 8 - 出力ファイルを PC として Natural に定義する場合のレポート指定

```
** Example 'PCDIEX1': DISPLAY and WRITE to PC
**
** NOTE: Example requires that Natural Connection is installed.
*****
DEFINE DATA LOCAL
01 PERS VIEW OF EMPLOYEES
  02 PERSONNEL-ID
  02 NAME
  02 CITY
END-DEFINE
*
FIND PERS WITH CITY = 'NEW YORK'           /* Data selection
  WRITE (7) TITLE LEFT 'List of employees in New York' /
  DISPLAY (7)                               /* (7) designates the output file (here the PC).
    'Location' CITY
    'Surname'   NAME
    'ID'       PERSONNEL-ID
END-FIND
END
```



# 54 DIVIDE

---

■ 機能 .....	342
■ 構文説明 .....	342
■ 例 .....	345

関連ステートメント：[ADD](#) | [COMPRESS](#) | [COMPUTE](#) | [EXAMINE](#) | [MOVE](#) | [MOVE ALL](#) | [MULTIPLY](#) | [RESET](#) | [SEPARATE](#) | [SUBTRACT](#)

関連機能グループ：「[算術演算とデータ移動操作](#)」

## 機能

DIVIDE ステートメントは、2つのオペランドを除算するために使用します。



**注意:** ゼロによる除算に関する注意事項：ゼロになっている除数 (*operand1*) を使用しようとすると、エラーメッセージまたは結果としてゼロが返されます。いずれが返されるかは、セッションパラメータ ZD の設定によって決まります（詳細については、『[パラメータリファレンス](#)』ドキュメントを参照）。

## 構文説明

このステートメントには異なる構造が可能です。

- [構文 1 - DIVIDE \(GIVING 節を含まない場合\)](#)
- [構文 2 - DIVIDE \(GIVING 節を含む場合\)](#)
- [構文 3 - DIVIDE \(REMAINDER オプションを含む場合\)](#)

### 構文 1 - DIVIDE (GIVING 節を含まない場合)

```
DIVIDE [ROUNDED] operand1 INTO operand2
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S A	N N P I F	可	不可
<i>operand2</i>	C S A	M N P I F	可	不可



構文要素の説明：

<i>operand1</i> INTO <i>operand2</i>	<p>オペランド：</p> <p><i>operand1</i>は除数で、<i>operand2</i>は被除数です。結果は <i>operand2</i> (結果フィールド) に格納されるため、ステートメントは次の式と同等になります。</p> <pre>&lt;oper2&gt; := &lt;oper2&gt; / &lt;oper1&gt;</pre> <p>結果フィールドとして使用できるのは、データベースフィールドまたはユーザー定義変数です。<i>operand2</i>が定数または変更不可のNaturalシステム変数の場合は、GIVING節が必要です。除算の結果の小数点位置は、結果フィールド (つまり、<i>operand2</i>) から計算されます。</p>
ROUNDED	キーワード ROUNDED を指定すると、結果は切り上げられます。

## 構文 2 - DIVIDE (GIVING 節を含む場合)

```
DIVIDE [ROUNDED] operand1 INTO operand2 [GIVING operand3]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S A N	N P I F	可	不可
<i>operand2</i>	C S A N	N P I F	可	不可
<i>operand3</i>	S A	A U N P I F B*	可	可

\**operand3* のフォーマット B は、4 以下の長さでのみ使用できます。

構文要素の説明：

<i>operand1</i> INTO <i>operand2</i> GIVING <i>operand3</i>	オペランド：
--	--------

	<p><i>operand1</i> は除数で、<i>operand2</i> は被除数です。結果は <i>operand3</i> に格納されるため、ステートメントは次の式と同等になります。</p> <pre>&lt;oper3&gt; := &lt;oper2&gt; / &lt;oper1&gt;</pre> <p>結果フィールドとしてデータベースフィールドを使用している場合は、除算の結果により更新されるのはプログラム内で使用されるフィールドの内部値のみです。データベース内のフィールドの値は変更されないまま維持されます。</p> <p>除算の結果の小数点位置は、結果フィールド（つまり、<i>operand3</i>）から計算されます。</p> <p>結果の精度については、『プログラミングガイド』の「演算割り当てのルール」、「算術演算結果の精度」を参照してください。</p>
<b>ROUNDED</b>	<p>キーワード <b>ROUNDED</b> を指定すると、結果は切り上げられます。</p>

### 構文 3 - DIVIDE (REMAINDER オプションを含む場合)

```
DIVIDE operand1 INTO operand2 [GIVING operand3] REMAINDER operand4
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S A	N N P I	可	不可
<i>operand2</i>	C S A	N N P I	可	不可
<i>operand3</i>	S A	A U N P I F B*	可	可
<i>operand4</i>	S A	A U N P I F B* T	可	可

\* *operand3* および *operand4* のフォーマット B は、4 以下の長さでのみ使用できます。

構文要素の説明：

<i>operand1</i>	<p>除数：</p> <p><i>operand1</i> は除数、つまり割り算の商を求めるために被除数を割る数値または量のことです。</p>
<i>operand2</i>	<p>結果フィールド：</p> <p><b>GIVING</b> 節を使用しない場合は、<i>operand2</i> に結果が格納されます。結果フィールドとして使用できるのは、データベースフィールドまたはユーザー定義変数です。</p>

	<i>operand2</i> が定数または変更不可の Natural システム変数の場合は、GIVING 節が必要です。
<b>ROUNDED</b>	キーワード ROUNDED を指定すると、結果は切り上げられます。
<b>GIVING</b> <i>operand3</i>	<p>キーワード GIVING を使用すると、<i>operand2</i> は変更されず、<i>operand3</i> に結果が格納されます。</p> <p>結果フィールドとしてデータベースフィールドを使用している場合は、除算の結果により更新されるのはプログラム内で使用されるフィールドの内部値のみです。データベース内のフィールドの値は変更されないまま維持されます。</p> <p>除算結果の小数桁数は、結果フィールド（つまり GIVING 節を使用している場合は <i>operand2</i>、GIVING 節を使用していない場合は <i>operand3</i>）で評価されます。</p> <p>結果の精度については、『プログラミングガイド』の「演算割り当てのルール」、「算術演算結果の精度」を参照してください。</p>
<b>REMAINDER</b> <i>operand4</i>	<p>キーワード REMAINDER を指定すると、除算の余りが指定フィールド（<i>operand4</i>）に挿入されます。</p> <p>GIVING と REMAINDER を併用する場合、4つのオペランドに配列範囲を使用することはできません。</p> <p>余りは、内部的に次のように計算されます。</p> <ol style="list-style-type: none"> <li>1. <i>operand1</i> と <i>operand2</i> の除算の商が計算されます。</li> <li>2. 商は <i>operand1</i> と掛け合わされます。</li> <li>3. この乗算の結果が <i>operand2</i> から引かれます。</li> <li>4. この減算の結果は、<i>operand4</i> に割り当てられます。</li> </ol> <p>これらの各ステップにおいて、「算術演算結果の精度」（『プログラミングガイド』）で説明されている規則が適用されます。</p>

## 例

```

** Example 'DIVEX1': DIVIDE
*****
DEFINE DATA LOCAL
1 #A (N7) INIT <20>
1 #B (N7)
1 #C (N3.2)
1 #D (N1)
1 #E (N1) INIT <3>
1 #F (N1)
END-DEFINE
*
DIVIDE 5 INTO #A
WRITE NOTITLE 'DIVIDE 5 INTO #A' 20X '=' #A

```

## DIVIDE

---

```
*
RESET INITIAL #A
DIVIDE 5 INTO #A GIVING #B
WRITE 'DIVIDE 5 INTO #A GIVING #B' 10X '=' #B
*
DIVIDE 3 INTO 3.10 GIVING #C
WRITE 'DIVIDE 3 INTO 3.10 GIVING #C' 8X '=' #C
*
DIVIDE 3 INTO 3.1 GIVING #D
WRITE 'DIVIDE 3 INTO 3.1 GIVING #D' 9X '=' #D
*
DIVIDE 2 INTO #E REMAINDER #F
WRITE 'DIVIDE 2 INTO #E REMAINDER #F' 7X '=' #E '=' #F
*
END
```

プログラム **DIVEX1** の出力：

```
DIVIDE 5 INTO #A           #A:      4
DIVIDE 5 INTO #A GIVING #B #B:      4
DIVIDE 3 INTO 3.10 GIVING #C #C:    1.03
DIVIDE 3 INTO 3.1 GIVING #D #D:     1
DIVIDE 2 INTO #E REMAINDER #F #E:   1 #F:  1
```

# 55 DO/DOEND

---

▪ 機能 .....	348
▪ 制限事項 .....	348
▪ 例 .....	349

```
DO statement ... DOEND
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

---

DO および DOEND ステートメントは、次のステートメントの論理条件に基づいて実行されるステートメント群を指定するために使用します。

- AT BREAK
- AT END OF DATA
- AT END OF PAGE
- AT START OF DATA
- AT TOP OF PAGE
- BEFORE BREAK PROCESSING
- FIND ... IF NO RECORDS FOUND
- IF
- IF SELECTION
- ON ERROR
- READ WORK FILE ... AT END OF FILE

## 制限事項

---

- DO および DOEND ステートメントはレポーティングモードでのみ有効です。
- WRITE TITLE、WRITE TRAILER、および AT 条件ステートメント AT BREAK、AT END OF DATA、AT END OF PAGE、AT START OF DATA、AT TOP OF PAGE は、DO/DOEND ステートメント群の中では指定できません。
- ループ開始ステートメントを DO/DOEND 内のステートメント群に使用する場合、そのループは DOEND の前に LOOP ステートメントで終了する必要があります。

## 例

```

** Example 'DOEEX1': DO/DOEND
*****
*
EMP. FIND EMPLOYEES WITH CITY = 'MILWAUKEE'
  VEH. FIND VEHICLES WITH PERSONNEL-ID = PERSONNEL-ID
    IF NO RECORDS FOUND DO
      ESCAPE
    DOEND
    DISPLAY PERSONNEL-ID (EMP.) NAME (EMP.)
      SALARY (EMP.,1)
      MAKE (VEH.) MAINT-COST (VEH.,1)
  AT END OF DATA DO
    WRITE NOTITLE
      / 10X 'AVG SALARY:'
        T*SALARY (1) AVER(SALARY (1))
      / 10X 'AVG MAINTENANCE (ZERO VALUES EXCLUDED):'
        T*MAINT-COST (1) NAVER(MAINT-COST (1))
    DOEND
  /*
LOOP
LOOP
END

```

プログラム **DOEEX1** の出力：

PERSONNEL ID	NAME	ANNUAL SALARY	MAKE	MAINT-COST
20021100	JONES	31000	GENERAL MOTORS	140
20027800	LAWLER	29000	GENERAL MOTORS	0
20027800	LAWLER	29000	TOYOTA	86
20030600	NORDYKE	47000	FORD	194
	AVG SALARY:	35666		
	AVG MAINTENANCE (ZERO VALUES EXCLUDED):			140






# 56

## DOWNLOAD PC FILE

---

■ 機能 .....	352
■ 構文説明 .....	352
■ 例 .....	353



構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CLOSE PC FILE](#) | [UPLOAD PC FILE](#) | [WRITE WORK FILE](#)

関連機能グループ：「[ワークファイル/PC ファイルの制御](#)」

## 機能

このステートメントは、データをメインフレーム プラットフォームから PC に転送するために使用します。

詳細については、『[Natural Connection](#)』ドキュメントと『[Entire Connection](#)』ドキュメントを参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
operand1	C S A G	A U N P I F B D T L C	可	不可
operand2	C S	A	可	可

フォーマット C と G は、いずれも [Natural Connection](#) では無効です。

構文要素の説明：

<i>work-file-number</i>	使用するワークファイル番号を指定します。この番号は、 <a href="#">Natural</a> に対する定義に従って、PC のワークファイル番号の 1 つに対応している必要があります。
<b>VARIABLE</b>	PC ファイルに書き込まれるレコードは可変フォーマットです。可変レコードは PC のスプレッドシート形式に変換できません。
<i>operand1</i>	フィールド指定：  <i>operand1</i> では、PC にダウンロードするフィールドを指定します。
<b>COMMAND</b>	COMMAND 節を使用すると、メインフレームから PC に PC コマンド（つまり、PC 上の <a href="#">Entire Connection</a> のコマンド行に入力できる任意のコマンド）を送信できます。

	<p>Entire Connection 側で、送信されたコマンドが有効かどうかチェックされます。PC で認識できないコマンドは拒否されます。この場合は、ダウンロードされたコマンドが PC で拒否されたことを示すエラーメッセージが発行されます。</p> <p>COMMAND 節を使用すると、例えば、メインフレームから Entire Connection のタスクを実行できます。PC のディレクトリ情報を一覧表示するタスク DIR を実行する場合は、メインフレーム上の Natural プログラムで次のステートメントを使用してすぐに実行できます。</p> <pre>DOWNLOAD PC FILE 7 COMMAND 'DIR'</pre> <p>下記の「例2」では、COMMAND 節を使用して、ダウンロードされたデータを受け取る PC ファイルの名前を定義します。この方法では、ファイル名を入力する手間を省くことができます。</p>
<i>operand2</i>	<p><b>COMMAND</b> の指定：</p> <p><i>operand2</i> では、PC 上で実行する DOS コマンドまたは Entire Connection タスクを指定します。 <i>operand2</i> は、英数字の定数または変数にする必要があります。</p>
<b>SYNC</b>	<p>SYNC では、COMMAND の実行と終了後に PC から Natural に制御を返すことを指定します。 SYNC を使用すると、例えば、ファイル転送を開始する前にコマンド SET PCFILE を確実に実行することができます。</p>
<b>ASYNC</b>	<p>ASYNC では、COMMAND の実行が終了したかどうかに関係なく、PC から Natural に制御を直ちに返すことを指定します。</p>

## 例

- [例 1 - DOWNLOAD PC FILE ステートメントの使用例](#)
- [例 2 - COMMAND 節の使用例](#)

### 例 1 - DOWNLOAD PC FILE ステートメントの使用例

次のプログラムで、DOWNLOAD PC FILE ステートメントの使用方法を示します。最初にデータが選択され、続いて Work File 7 を使用して PC にダウンロードされます。

```
** Example 'PCDOEX1': DOWNLOAD PC FILE
**
** NOTE: Example requires that Natural Connection is installed.
*****
DEFINE DATA LOCAL
01 PERS VIEW OF EMPLOYEES
   02 PERSONNEL-ID
   02 NAME
   02 CITY
END-DEFINE
*
```

```
FIND PERS WITH CITY = 'NEW YORK'          /* Data selection
  DOWNLOAD PC FILE 7 CITY NAME PERSONNEL-ID /* Data download
END-FIND
END
```

プログラム **PCDOEX1** の出力：

プログラムを実行するとウィンドウが開くため、データがダウンロードされるPCファイルの名前を指定します。その後、データがPCにダウンロードされます。

## 例 2 - COMMAND 節の使用例

次のプログラムで、DOWNLOAD PC FILE ステートメントでの COMMAND 節の使用方法を示します。最初に、データを受け取る側のPCファイルの名前が定義されます。次に、データが選択され、このファイルにダウンロードされます。

```
** Example 'PCDOEX2': DOWNLOAD PC FILE
**
** NOTE: Example requires that Natural Connection is installed.
*****
DEFINE DATA LOCAL
01 PERS VIEW OF EMPLOYEES
  02 PERSONNEL-ID
  02 NAME
  02 CITY
01 CMD (A80)          /* Variable for transfer
END-DEFINE          /* of the PC command
*
MOVE 'SET PCFILE 7 DOWN DATA PERS.NCD' TO CMD /* PC command to define
*
DOWNLOAD PC FILE 6 COMMAND CMD          /* Command download
*
FIND PERS WITH CITY = 'NEW YORK'      /* Data selection
  DOWNLOAD PC FILE 7 CITY NAME PERSONNEL-ID /* Data download
END-FIND
END
```



**注意:** 2つの連続した DOWNLOAD PC FILE ステートメントでは、PCファイルの番号は異なっている必要があります。

プログラム **PCDOEX2** の出力：

プログラムを実行すると、プログラム内で指定したPCファイルにデータがダウンロードされます。ウィンドウは表示されません。

# 57 EJECT

---

▪ 機能 .....	356
▪ 構文説明 .....	356
▪ 処理 .....	358
▪ 例 .....	358

関連ステートメント：[AT END OF PAGE](#) | [AT TOP OF PAGE](#) | [CLOSE PRINTER](#) | [DEFINE PRINTER](#) | [DISPLAY](#) | [FORMAT](#) | [NEWPAGE](#) | [PRINT](#) | [SKIP](#) | [SUSPEND IDENTICAL SUPPRESS](#) | [WRITE](#) | [WRITE TITLE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

EJECT ステートメントは、改ページとページ換えの制御に使用します。

## 構文説明

このステートメントには、2つの異なる構造が可能です。

- [EJECT - 構文 1](#)
- [EJECT - 構文 2](#)

次の構文図で使用されている記号については、「[構文記号](#)」を参照してください。

### EJECT - 構文 1

```
EJECT { ON }
      { OFF } [(rep)]
```

構文要素の説明：

<b>EJECT ON/OFF</b> (rep)	レポート指定を含む <b>EJECT ON/OFF</b> - オンラインおよびバッチモード：	
	EJECT OFF (rep)	指定レポートのページ換え処理を行いません（ <a href="#">構文 2</a> の EJECT ステートメントの指定は除く）。
	EJECT ON (rep)	指定レポートのページ換え処理を実行します。
<b>EJECT ON/OFF</b>	レポート指定を含まない <b>EJECT ON/OFF</b> - バッチモードのみ：	
	レポート指定 (rep) のない EJECT ON/OFF では、バッチモードで使用して、プログラムの実行中に作成される出力リスト間のページ換え処理を制御できます。	
	EJECT ON	ソースプログラムリスト、出力レポートおよび実行完了メッセージ「EXECUTION COMPLETED」の間でページ換えを実行します。これはデフォルト設定です。
	EJECT OFF	上記の各出力間のページ換えを停止します。EJECT OFF は次の EJECT ON ステートメントが指定されるまで有効です。

<i>(rep)</i>	<p>レポート指定：</p> <p>表記 (<i>rep</i>) は、EJECT ステートメントを適用するレポートの ID を指定するために使用できません。</p> <p>範囲 0~31 の値、または <b>DEFINE PRINTER</b> ステートメントを使用して割り当てた論理名を指定できます。</p> <p>(<i>rep</i>) を指定しない場合、EJECT ステートメントは最初のレポート（レポート 0）に適用されます。</p> <p>Natural で作成される出力レポートの形式を制御する方法については、『プログラミングガイド』の「データ出力制御」を参照してください。</p>
--------------	---

## EJECT - 構文 2

この形式の EJECT ステートメントでは、次のページでタイトルまたはヘッダ行を生成することなく、TOP/END PAGE 処理を行わずに改ページが実行されます。

```
EJECT [(rep)] [ [ IF ] LESS [THAN] operand1 [LINES] [LEFT] ]
                [ [ WHEN ] ]
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S	N P I	可	不可

構文要素の説明：

<i>(rep)</i>	<p>レポート指定：</p> <p>表記 (<i>rep</i>) は、EJECT ステートメントを適用するレポートの ID を指定するために使用できません。</p> <p>範囲 0~31 の値、または <b>DEFINE PRINTER</b> ステートメントを使用して割り当てた論理名を指定できます。</p> <p>(<i>rep</i>) を指定しない場合、EJECT ステートメントは最初のレポート（レポート 0）に適用されます。</p> <p>Natural で作成される出力レポートの形式を制御する方法については、『プログラミングガイド』の「データ出力制御」を参照してください。</p>
<b>IF LESS THAN</b> <i>operand1</i> <b>LINES</b> <b>LEFT</b>	<p>現在の行の位置が、ページサイズから <i>operand1</i> の値を引いた値よりも大きいときに限り、ページ換えが行われます。 <i>operand1</i> の値は数値定数または変数で指定できます。</p>

## 処理

EJECT ステートメントが実行されても、AT TOP OF PAGE、AT END OF PAGE、WRITE TITLE、または WRITE TRAILER ステートメントが実行されることはありません。また、DISPLAY GIVE SYSTEM FUNCTIONS で評価されるシステム関数にも影響しません。

EJECT では、新規の物理ページの生成だけが行われます。Natural システム変数 \*LINE-COUNT が 1 に設定されますが、\*PAGE-NUMBER は影響を受けません。

## 例

```

** Example 'EJTEX1': EJECT
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 NAME
  2 JOB-TITLE
END-DEFINE
*
FORMAT PS=15
LIMIT 9
READ EMPLOY-VIEW BY CITY
/*
  AT START OF DATA
    EJECT
    WRITE /// 20T '%' (29) /
              20T '%%'                               47T '%%' /
              20T '%%' 3X 'REPORT OF EMPLOYEES' 47T '%%' /
              20T '%%' 3X ' SORTED BY CITY   ' 47T '%%' /
              20T '%%'                               47T '%%' /
              20T '%' (29) /

    EJECT
  END-START
  EJECT WHEN LESS THAN 3 LINES LEFT
/*
  WRITE '*' (64)
  DISPLAY NOTITLE NOHDR CITY NAME JOB-TITLE 5X *LINE-COUNT
  WRITE '*' (64)
END-READ
END

```



プログラム EJTEX1 の出力：

```

%%%%%%%%%%
%%
%% REPORT OF EMPLOYEES %%
%% SORTED BY CITY %%
%%
%%
%%%%%%%%%%

```

Enter キーを押した後：

```

*****
AIKEN          SENKO          PROGRAMMER          2
*****
AIX EN OTHE    GODEFROY          COMPTABLE          5
*****
AJACCIO        CANALE            CONSULTANT          8
*****
ALBERTSLUND    PLOUG             KONTORASSISTENT    11
*****
ALBUQUERQUE    HAMMOND           SECRETARY           14
*****

```

Enter キーを押した後：

```

*****
ALBUQUERQUE    ROLLING           MANAGER             2
*****
ALBUQUERQUE    FREEMAN           MANAGER             5
*****
ALBUQUERQUE    LINCOLN           ANALYST             8
*****
ALFRETON        GOLDBERG          JUNIOR              11
*****


```



# 58 END

---

■ 機能 .....	362
■ 構文説明 .....	362
■ 例 .....	363



構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

END ステートメントは、Natural プログラムの物理的な終わりをマークするために使用します。END ステートメントの後に記号を挿入することはできません。

レポーティングモードでは、現在アクティブになっているすべての処理ループ（LOOP ステートメントで終了していないもの）が END ステートメントで終了します。

### プログラム実行に関する考慮事項

メインプログラム（レベル 1 で実行中のプログラム）で END ステートメントに到達すると、参照指定 (*r*) で処理ループに関連付けられていないユーザー開始ブレイク（[PERFORM BREAK PROCESSING](#)）の最終ブレイク処理と、最後の END PAGE 処理が行われます。

サブプログラムまたは [FETCH RETURN](#) ステートメントで呼び出されたプログラムで、END ステートメントに到達すると、終了処理を行わずに制御が呼び出し元のプログラムに返されます。

## 構文説明

END	<p>キーワード：</p> <p>通常、Natural 予約キーワード END ステートメントは、Natural プログラムの物理的な終わりをマークするために使用します。</p>
.	<p>ピリオド：</p> <p>Natural 予約キーワード END の代わりに、ピリオド (.) を使用することもできます。同じ行に他のステートメントが含まれている場合は、ピリオドの前に少なくとも 1 つの空白を挿入する必要があります。</p>

## 例

---

いくつかの一般的な例については、「[DEFINE DATA ステートメントの使用例](#)」を参照してください。



# 59

## END TRANSACTION

---

■ 機能 .....	366
■ 制限事項 .....	366
■ 構文説明 .....	367
■ 関連データベース .....	367
■ データベース固有の考慮事項 .....	368
■ 例 .....	368

END [OF] TRANSACTION [*operand1* ...]

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION DATA](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

---

END TRANSACTION ステートメントは、論理トランザクションの終了を示すために使用します。論理トランザクションは、データベースに含まれている情報の論理的な一貫性を保証するために、完全に実行される必要がある最小の作業論理ユニットです。この定義はユーザーが行います。

END TRANSACTION ステートメントが正しく実行されると、後続のユーザー、Natural、データベース、またはオペレーティングシステムによる中断に関係なく、トランザクション中のすべての更新処理をデータベースに対して物理的に適用することが保証されます。END TRANSACTION ステートメントが正しく終了しなかった場合、トランザクション中の更新処理は自動的にバックアウトされます。

また、END TRANSACTION ステートメントでは、トランザクション中にホールド状態となっていたすべてのレコードが解放されます。

END TRANSACTION ステートメントは、論理条件に基づいて実行できます。

詳細については、『[プログラミングガイド](#)』の「[データベース更新-トランザクション処理](#)」セクションを参照してください。

## 制限事項

---

このステートメントを、Entire System Server で使用することはできません。



## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	C S   N A U N P I F B D T		可	不可

構文要素の説明：

<i>operand1</i>	<p>トランザクションデータの保存：</p> <p>Adabas データベース、またはバッチ関連 BMP リージョン内の DL/I データベース（IMS 環境のみ）に適用されるトランザクションに対して、ユーザーは、このステートメントを使用して、トランザクション関連情報を保存することもできます。ただし、トランザクションデータは 2000 バイトを超えないようにする必要があります。トランザクションデータは、<b>GET TRANSACTION DATA</b> ステートメントで読み込むことができます。</p> <p>トランザクションデータは、Natural プロファイルパラメータ ETDB で指定したデータベースに書き込まれます。</p> <p>ETDB パラメータを指定していない場合、トランザクションデータは、プロファイルパラメータ UDB で指定したデータベースに書き込まれます（メインフレーム環境を除く）。メインフレーム環境では、トランザクションデータは、Natural Security システムファイル（FSEC）が存在するデータベースに書き込まれます。FSEC を指定していない場合、Natural システムファイル（FNAT）と同じであるとみなされます。Natural Security をインストールしていない場合は、トランザクションデータは FNAT が存在するデータベースに書き込まれます。</p>
-----------------	--

## 関連データベース

トランザクションデータ (*operand1*) が指定されていない END TRANSACTION ステートメントは、Natural の制御下で、データベーストランザクションが行われたときのみ実行されます。Natural プロファイルパラメータ ET の設定により、ステートメントがどのように実行されるかが決まります。ET=OFF を指定すると、トランザクションの影響を受けたデータベースに対してのみ実行されます。ET=ON を指定すると、最後の **BACKOUT TRANSACTION** または END TRANSACTION ステートメントの実行以後に参照された、全データベースに対して実行されます。

トランザクションデータ (*operand1*) が指定されている END TRANSACTION ステートメントは常に実行され、データベースにトランザクションデータが格納されます（下記のセクションを参照）。また、他のどのデータベースに対して END TRANSACTION ステートメントが実行されるかは、ET パラメータの設定で決まります（上記を参照）。

## データベース固有の考慮事項

DLI データベース	PSB スケジューリングはシンクポイント要求によって終了するため、Natural では、END TRANSACTION ステートメントの実行前に PSB ポジションが保存されます。次のコマンドを実行する前に、Natural では、PSB の再スケジュールを行って PCB ポジションを END TRANSACTION ステートメント前のポジションに設定することが試みられます。ポイントされているセグメントが END TRANSACTION ステートメントと次のコマンドの間で削除されていた場合は、PCB ポジションが前方にシフトされることもあります。
VSAM データベース	VSAM にアクセスするときに適用されるトランザクションロジックについては、『データベース管理システムインターフェイス』ドキュメントの「Natural for VSAM」を参照してください。
SQL データベース	ほとんどの SQL データベースは作業の論理ユニットが終了するときにすべてのカーソルを閉じるため、END TRANSACTION ステートメントをデータベースの更新処理ループ内に指定することはできません。このようなループの後に指定する必要があります。

## 例

- 例 1 - END TRANSACTION
- 例 2 - ET データを指定した END TRANSACTION

### 例 1 - END TRANSACTION

```

** Example 'ETREX1': END TRANSACTION
**
** CAUTION: Executing this example will modify the database records!
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 COUNTRY
END-DEFINE
*
FIND EMPLOY-VIEW WITH CITY = 'BOSTON'
  ASSIGN COUNTRY = 'USA'
  UPDATE
  END TRANSACTION
/*
AT END OF DATA
  WRITE NOTITLE *NUMBER 'RECORDS UPDATED'
END-ENDDATA
/*
END-FIND
END

```

プログラム **ETREX1** の出力：

```
7 RECORDS UPDATED
```

## 例 2 - ET データを指定した END TRANSACTION

```
** Example 'ETREX2': END TRANSACTION (with ET data)
**
** CAUTION: Executing this example will modify the database records!
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
  2 CITY
*
1 #PERS-NR (A8) INIT <' '>
END-DEFINE
*
REPEAT
  INPUT 'ENTER PERSONNEL NUMBER TO BE UPDATED:' #PERS-NR
  IF #PERS-NR = ' '
    ESCAPE BOTTOM
  END-IF
  /*
  FIND EMPLOY-VIEW PERSONNEL-ID = #PERS-NR
  INPUT (AD=M)   NAME / FIRST-NAME / CITY
  UPDATE
  END TRANSACTION #PERS-NR
  END-FIND
  /*
END-REPEAT
END
```

プログラム **ETREX2** の出力：

```
ENTER PERSONNEL NUMBER TO BE UPDATED: 20027800
```

社員番号を入力して確認した後：

```
NAME LAWLER
FIRST-NAME SUNNY
CITY MILWAUKEE
```



# 60 ESCAPE

---

▪ 機能 .....	372
▪ 構文説明 .....	373
▪ 例 .....	374

## ストラクチャードモード構文

ESCAPE	}	TOP [REPOSITION]
		BOTTOM [(r)] [IMMEDIATE]
		ROUTINE [IMMEDIATE]
		MODULE [IMMEDIATE]

## レポートモード構文

ESCAPE	}	}	TOP [REPOSITION]
			BOTTOM [(r)] [IMMEDIATE]
			ROUTINE [IMMEDIATE]
			MODULE [IMMEDIATE]

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：

- [FOR](#) | [REPEAT](#) | [PROCESS PAGE MODAL](#)
- [CALL](#) | [CALL FILE](#) | [CALL LOOP](#) | [CALLNAT](#) | [DEFINE SUBROUTINE](#) | [FETCH](#) | [PERFORM](#)

関連機能グループ：

- [ループ実行](#)
- [プログラムおよびルーチンの呼び出し](#)

## 機能

ESCAPE ステートメントは、処理ループやルーチンの実行を中断するために使用します。

キーワードの TOP、BOTTOM、および ROUTINE によって、ESCAPE ステートメントの実行後に処理をどこから続けるかを指定します。

ESCAPE TOP および ESCAPE BOTTOM ステートメントは、内部的に最も内側でアクティブになっている処理ループを参照します。ESCAPE ステートメントを、物理的に処理ループ内に置く必要はありません。

ESCAPE TOP および ESCAPE BOTTOM ステートメントをルーチン内（サブルーチン、サブプログラム、または FETCH RETURN で起動されたプログラム）に指定すると、処理ループの実行中に呼び出されたルーチンは自動的に終了します。

## その他の考慮事項

同一の処理ループ内に、複数の ESCAPE ステートメントを指定できます。

ESCAPE ステートメントは、論理条件に基づいて実行できます。ESCAPE ステートメントが AT END OF DATA、AT BREAK、または AT END OF PAGE ブロックの処理中に実行されると、特殊条件ブロックの実行を終了し、指定された ESCAPE 処理を行います。

IF NO RECORDS FOUND 条件の処理中に ESCAPE ステートメントが実行されると、ループ終了処理は行われません (ESCAPE IMMEDIATE と同じ)。

## 構文説明

<b>ESCAPE TOP</b>	ESCAPE TOP を実行すると、処理ループの先頭から処理が続行されます。処理ループが再び先頭から繰り返されます。
<b>REPOSITION</b>	ESCAPE TOP REPOSITION ステートメントを実行すると、Natural では開始値として検索変数の現在の値を使用して、即座にアクティブな READ ループの先頭から処理が続行されます。  同時に、ESCAPE TOP REPOSITION により、システム変数 *COUNTER がゼロにリセットされます。  ESCAPE TOP REPOSITION は、Adabas、DL/I、または VSAM データベースにアクセスする READ ステートメント処理ループ内で指定できます。関連する READ ステートメントでは、WITH REPOSITION オプションを指定する必要があります。
<b>ESCAPE BOTTOM</b>	ESCAPE BOTTOM を実行すると、処理ループ終了後の最初のステートメントから処理が続行されます。ループは終了し、ループ終了処理 (最終の BREAK 処理、END DATA 処理) が実行されます。  レポートモードでは、ESCAPE BOTTOM がデフォルトです。
<i>(r)</i>	表記 <i>(r)</i> : ESCAPE BOTTOM に続けてラベルや参照番号を指定すると、ラベルや参照番号で示した処理ループの直後の最初のステートメントから処理が続行されます。  ラベルまたは参照番号を指定できるのは、参照する処理ループに ESCAPE BOTTOM ステートメントが物理的に挿入されている場合のみです。
<b>IMMEDIATE</b>	キーワード IMMEDIATE を指定すると、ループ終了処理は行われません。
<b>ESCAPE ROUTINE</b>	ESCAPE ROUTINE を実行すると、PERFORM、CALLNAT、FETCH RETURN で呼び出された現在の Natural ルーチン、またはメインプログラムとしての現在の Natural ルーチンから制御が解放されます。  サブルーチンの場合、そのサブルーチンを呼び出したステートメントの、次のステートメントから処理が続行されます。メインプログラムの場合は、Natural コマンドモードに入ります。  ルーチン内でアクティブになっているループはすべて終了し、ループ終了処理、およびユーザー開始 (PERFORM BREAK) 処理に対する最終処理が行われます。ESCAPE ROUTINE を含

	むプログラムをメインプログラム（レベル1）として実行している場合に、最終のENDPAGE処理が行われます。
<b>ESCAPE MODULE</b>	<p>ESCAPE MODULE を実行すると、現在のプログラムレベル全体（すべての内部サブルーチンを含む）の制御が中止されます。その後、制御は前のプログラムレベルのオブジェクトに戻されます。内部サブルーチンの階層内で ESCAPE MODULE を使用すると、このレベルで動作しているすべてのルーチンを同時に終了できます。アクティブな内部サブルーチンがない場合、ESCAPE MODULE の結果は ESCAPE ROUTINE と同じになります。</p> <p>ESCAPE MODULE は、内部サブルーチンにのみに関係しています。外部サブルーチン、サブプログラム、および呼び出されたプログラムでは、ESCAPE ROUTINE と同じ効果がありません。</p> <p>ESCAPE ROUTINE と同様に、ループ終了処理が実行されます。ただし、キーワード IMMEDIATE を指定した場合、ループ終了処理は実行されません。</p>

## 例

```

** Example 'ESCEX1': ESCAPE
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 FIRST-NAME
  2 NAME
  2 AREA-CODE
  2 PHONE
*
1 #CITY (A20) INIT <' '>
1 #CNTL (A1) INIT <' '>
END-DEFINE
*
REPEAT
  INPUT 'ENTER VALUE FOR CITY: ' #CITY
    / 'OR ''.' TO TERMINATE '
  IF #CITY = '.'
    ESCAPE BOTTOM
  END-IF
/*
FND. FIND EMPLOY-VIEW WITH CITY = #CITY
/*
  IF NO RECORDS FOUND
    WRITE 'NO RECORDS FOUND'
    ESCAPE BOTTOM (FND.)
  END-NOREC
AT START OF DATA
  INPUT (AD=0) 'RECORDS FOUND:' *NUMBER //
    'ENTER ''D'' TO DISPLAY RECORDS' #CNTL (AD=M)
  IF #CNTL NE 'D'

```



```
    ESCAPE BOTTOM (FND.)
    END-IF
    END-START
    /*
    DISPLAY NOTITLE NAME FIRST-NAME PHONE
    END-FIND
    END-REPEAT
```

プログラム **ESCEX1** の出力：

```
ENTER VALUE FOR CITY:  PARIS
(OR '.' TO TERMINATE)
```

市町村名を入力して確認した後：

```
RECORDS FOUND:          26
ENTER 'D' TO DISPLAY RECORDS D
```

**D** を入力して確認した後：

NAME	FIRST-NAME	TELEPHONE
MAIZIERE	ELISABETH	46758304
MARX	JEAN-MARIE	40738871
REIGNARD	JACQUELINE	48472153
RENAUD	MICHEL	46055008
REMOUE	GERMAINE	36929371
LAVENDA	SALOMON	40155905
BROUSSE	GUY	37502323
GIORDA	LOUIS	37497316
SIECA	FRANCOIS	40487413
CENSIER	BERNARD	38070268
DUC	JEAN-PAUL	38065261
CAHN	RAYMOND	43723961
MAZUY	ROBERT	44286899
FAURIE	HENRI	44341159
VALLY	ALAIN	47326249
BRETON	JEAN-MARIE	48467146
GIGLEUX	JACQUES	40477399
KORAB-BRZOZOWSKI	BOGDAN	45288048
XOLIN	CHRISTIAN	46060015
LEGRIS	ROGER	39341509
VVVV		



# 61 EXAMINE

---

▪ 構文 1 - EXAMINE .....	378
▪ 構文 2 - EXAMINE TRANSLATE .....	383
▪ 構文 3 - Unicode 書記素用の EXAMINE .....	384
▪ 例 .....	386

関連ステートメント：**ADD** | **COMPRESS** | **COMPUTE** | **DIVIDE** | **MOVE** | **MOVE ALL** | **MULTIPLY** | **RESET** | **SEPARATE** | **SUBTRACT**

関連機能グループ：「[算術演算とデータ移動操作](#)」

## 構文 1 - EXAMINE

```

EXAMINE
    [FULL [VALUE [OF]]] {
                                operand1
                                SUBSTRING
                                (operand1,operand2,operand3)
    }
    [FOR] [FULL [VALUE [OF]]] [PATTERN] operand4
    [DELIMITERS-option]
    {[DELETE-REPLACE-clause] [GIVING-clause]}
    
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

### 構文説明 - 構文 1

EXAMINE ステートメントは、英数字やバイナリフィールドの内容、または配列内にあるフィールドの範囲を調べるため、および次の目的で使用します。

- 検索パターンが見つかった回数を返します。
- 検索パターンが最初に出現したバイト位置を返します。
- フィールドの有効な内容の長さ（末尾の空白を除いたフィールド長）を返します。
- パターンが最初に見つかった配列フィールドのオカレンス番号（インデックス）を返します。
- パターンを別のパターンで置換します。
- パターンを削除します。

オペランド定義テーブル：

オペランド	構文要素		フォーマット										オペランド参照	ダイナミック定義					
<i>operand1</i>	C*	S A	A	U								B						可	不可
<i>operand2</i>	C	S					N	P	I	B*								可	不可
<i>operand3</i>	C	S					N	P	I	B*								可	不可
<i>operand4</i>	C	S		A	U							B						可	不可

\* GIVING 節を使用する場合、*operand1* に定数を指定できます。DELETE または REPLACE 節を使用する場合、定数の指定はできません。

\* *operand2* および *operand3* のフォーマット B は、4 以下の長さでのみ使用できます。

## 構文要素の説明：

<i>operand1</i>	<p><i>operand1</i> では、内容を調べるフィールドを指定します。</p> <p><i>operand1</i> が DYNAMIC 変数である場合は、REPLACE 操作でフィールドの長さを拡張または縮小できます。また、DELETE 操作で、フィールド長をゼロに設定することもできます。DYNAMIC 変数の現在の長さは、システム変数 *LENGTH を使用して確認できます。</p>
<i>operand4</i>	<i>operand4</i> では、EXAMINE 操作に使用する値を指定します。
<b>FULL</b>	オペランドに FULL を指定すると、末尾の空白も含めて値全体が処理されます。FULL を指定しない場合は、オペランド内の末尾の空白は無視されます。
<b>SUBSTRING</b>	<p>通常、フィールドの内容はフィールドの始めから終わりまでまたは空白以外の最後の文字まで調べられます。</p> <p>SUBSTRING オプションを使用すると、フィールドの特定の部分のみを調べることができます。SUBSTRING 節のフィールド名 (<i>operand1</i>) の後に、まず開始位置 (<i>operand2</i>)、次に調べるフィールド部分の長さ (<i>operand3</i>) を指定します。</p> <p>例えば、フィールド #A の 5 番目から 12 番目を調べるには、次のように指定します。</p> <pre>EXAMINE SUBSTRING(#A,5,8).</pre> <p><b>注意:</b></p> <ol style="list-style-type: none"> <li><i>operand2</i> を省略すると、開始位置は 1 と見なされます。</li> <li><i>operand3</i> を省略すると、長さはフィールドの開始位置から終わりまでと見なされます。</li> <li>SUBSTRING を DYNAMIC 変数と一緒に使用すると、フィールドは固定長変数と同様に動作します。つまり、DELETE または REPLACE 操作が実行されたかどうかに関係なく、EXAMINE 操作の結果として長さ (*LENGTH) が変更されることはありません。</li> </ol>
<b>PATTERN</b>	<p>「ワイルドカード文字」（調べる対象から除外する位置を示す記号）を含む値でフィールドを調べる場合は、PATTERN オプションを使用します。<i>operand4</i> では、無視する位置に次の記号を挿入できます。</p> <ul style="list-style-type: none"> <li>■ ピリオド (.)、疑問符 (?)、下線 ( _ ) は、調べる対象から除外する位置が単一の位置であることを示します。</li> <li>■ アスタリスク (*)、パーセント記号 (%) は、調べる対象から除外する位置の数がいくつでもよいことを示します。</li> </ul> <p>例：PATTERN 'NAT*AL' で、NAT と AL を含む、任意の値のフィールドを調べることができます。この場合、NAT と AL の間は、何文字でも許可されます。つまり、NATURAL や NATIONAL も NATAL と同じ結果になります。</p>

<i>DELIMITERS-option</i>	このオプションは、デリミタを示す値を検索するために使用します。詳細については、下記の「 <b>DELIMITERS オプション</b> 」を参照してください。
<i>DELETE-REPLACE-clause</i>	この節の DELETE オプションは、 <i>operand1</i> で見つかった各検索値 ( <i>operand4</i> ) を削除するために使用します。一方、REPLACE オプションは、 <i>operand6</i> で指定した値によって <i>operand1</i> で見つかった各検索値 ( <i>operand4</i> ) を置換するために使用します。詳細については、下記の「 <b>DELETE REPLACE 節</b> 」を参照してください。
<i>GIVING-clause</i>	詳細については、下記の「 <b>GIVING 節</b> 」を参照してください。

**DELIMITERS オプション**

```
{ ABSOLUTE
  [WITH DELIMITERS]
  [WITH DELIMITERS] operand5 }
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand5</i>	C S	A B	可	不可

構文要素の説明：

<b>ABSOLUTE</b>	これはデフォルトのオプションです。このオプションを指定すると、値が他のどんな文字に囲まれていても、フィールドに対して指定値の絶対検索が行われます。
<b>WITH DELIMITERS</b>	空白または英数字以外の文字で区切られた値を検索するために使用します。
<b>WITH DELIMITERS</b> <i>operand5</i>	<i>operand5</i> で指定した文字 (列) で区切られた値を検索するために使用します。

**DELETE-REPLACE 節**


```
[AND] { DELETE [FIRST]
        REPLACE [FIRST] [WITH] [FULL [VALUE [OF]]] operand6 }
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	動的定義
<i>operand6</i>	C S	A U B	可	不可


構文要素の説明：

<b>DELETE</b>	<i>operand1</i> の内容から、検索値 ( <i>operand4</i> ) の最初 (またはすべて) のオカレンスを削除するために使用します。
<b>REPLACE</b>	<i>operand1</i> にある検索値 ( <i>operand4</i> ) の最初 (またはすべて) のオカレンスを、 <i>operand6</i> で指定した値に置換します。
<b>FIRST</b>	キーワード FIRST を指定すると、最初の同じ値のみを削除または置換できます。

 **注意:**

1. REPLACE 操作の結果が *operand1* に入り切らない場合は、エラーメッセージが返されます。
2. *operand1* が動的変数である場合は、REPLACE 操作でフィールドの長さを拡張または縮小できます。また、DELETE 操作で、フィールド長をゼロに設定することもできます。動的変数の現在の長さは、システム変数 \*LENGTH を使用して確認できます。動的変数の一般的な情報については、「動的変数の使用」を参照してください。

**GIVING 節**



オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	動的定義
<i>operand7</i>	S	N P I	可	可

構文要素の説明：

<b>GIVING</b>	キーワード GIVING だけを指定した場合、これは GIVING NUMBER (デフォルト) に相当します。
<b>NUMBER</b>	内容を調べるフィールド ( <i>operand1</i> ) で検索値 ( <i>operand4</i> ) が見つかった回数を取得するために使用します。
<b>POSITION</b>	<i>operand1</i> (または <i>operand1</i> のサブstring) 内で <i>operand4</i> と同じ値が初めて出現したバイト位置を取得するために使用します。
<b>LENGTH</b>	すべての削除または置換操作が完了した後の <i>operand1</i> (または <i>operand1</i> のサブstring) に残っている内容の長さを取得するために使用します。末尾にある空白は無視されます。
<i>operand7</i>	検索値のオカレンス数を指定します。REPLACE FIRST または DELETE FIRST オプションを同時に使用する場合は、1 を超えないようにする必要があります。
<b>INDEX</b> <i>operand7...3</i>	下記を参照してください。

## GIVING INDEX

[GIVING] INDEX [IN] *operand7 ... 3*

このオプションは、調べられる基礎のフィールドが配列フィールドの場合にのみ適用できます。

構文要素の説明：

<b>INDEX</b>	GIVING INDEX は、最初の検索値 ( <i>operand4</i> ) が見つかった <i>operand1</i> の配列オカレンス番号 (インデックス) を取得するために使用します。
<i>operand7...3</i>	<i>operand1</i> の数 (最大 3 個) は、 <i>operand7</i> の次元数と一致させる必要があります。検索値がどのオカレンスでも見つからなかった場合、 <i>operand7</i> には 0 が入ります。



**注意:** *operand1* のインデックス範囲にオカレンス 0 (0:5 など) が含まれている場合、*operand7* の 0 値は正確ではありません。この場合、GIVING NUMBER 節を追加して検索値が実際に見つかったかどうかを明確にする必要があります。



## 構文 2 - EXAMINE TRANSLATE

EXAMINE	{	<i>operand1</i>	}	[AND]
		<u>SUBSTRING</u> ( <i>operand1,operand2,operand3</i> )		
TRANSLATE	{	INTO {	UPPER	[CASE]
			LOWER	}
		USING [INVERTED] ( <i>operand4</i> )		}

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

### 構文説明 - 構文 2

EXAMINE TRANSLATE ステートメントでは、フィールド内の文字を大文字や小文字、または他の文字に変換します。

オペランド定義テーブル：

オペランド	構文要素	フォーマット	オペランド参照	ダイナミック定義
<i>operand1</i>	S A	A B	可	不可
<i>operand2</i>	C S	N P I B*	可	不可
<i>operand3</i>	C S	N P I B*	可	不可
<i>operand4</i>	S A	A B	可	不可

\**operand2* および *operand3* のフォーマット B は、4 以下の長さでのみ使用できます。

構文要素の説明：

<b>EXAMINE</b> <i>operand1</i>	フィールド内容の全体の変換： <i>operand1</i> で指定したフィールドの内容が変換されます。
<b>EXAMINE SUBSTRING</b> <i>operand1 operand2</i> <i>operand3</i>	フィールド内容の一部分の変換： 通常は、フィールドの内容全体が変換されます。 SUBSTRING オプションを指定すると、フィールドの特定の部分のみが変換されます。SUBSTRING 節のフィールド名 ( <i>operand1</i> ) の後に、まず開始位置 ( <i>operand2</i> )、次に調べるフィールド部分の長さ ( <i>operand3</i> ) を指定します。

	<p>例えば、フィールド #A の 5 番目から 12 番目を変換する場合は、次のように指定します。</p> <pre>EXAMINE SUBSTRING(#A,5,8) AND TRANSLATE ...</pre> <p><b>注意:</b> <i>operand2</i> を省略すると、開始位置は 1 と見なされます。 <i>operand3</i> を省略すると、長さはフィールドの開始位置から終わりまでと見なされます。</p>
<b>TRANSLATE INTO UPPER CASE</b>	<p>大文字変換：</p> <p><i>operand1</i> の内容が大文字に変換されます。</p>
<b>TRANSLATE INTO LOWER CASE</b>	<p>小文字変換：</p> <p><i>operand1</i> の内容が小文字に変換されます。</p>
<b>TRANSLATE USING</b> <i>operand4</i>	<p>使用する変換テーブル：</p> <p><i>operand4</i> は、文字変換に使用する変換テーブルです。テーブルのフォーマット／長さは、A2 または B2 にする必要があります。</p> <p><b>注意:</b> 変換される文字に対して変換テーブルで複数の変換が定義されている場合は、最後の変換が適用されます。</p>
<b>INVERTED</b>	<p>キーワード <b>INVERTED</b> を指定すると、変換テーブル (<i>operand4</i>) は逆に使用されます。つまり、変換の方向が逆転します。</p>

### 構文 3 - Unicode 書記素用の EXAMINE

EXAMINE [FULL [VALUE [OF]]]	{	<i>operand1</i>	}
		SUBSTRING( <i>operand1</i> , <i>operand2</i> , <i>operand3</i> )	}
[FOR]	{	CHARPOSITION <i>operand4</i> CHARLENGTH <i>operand5</i>	}
		CHARPOSITION <i>operand4</i>	}
		CHARLENGTH <i>operand5</i>	}
[GIVING] POSITION IN <i>operand6</i> [[GIVING] LENGTH IN <i>operand7</i> ]			

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 構文説明 - 構文 3

「grapheme」は、ユーザーが一般に文字と見なすものです。ほとんどの場合、UTF-16 コード単位 (=U 形式文字) が書記素になりますが、書記素は複数のコード単位で構成されることもあります。例えば、一連の基本文字の後に、結合文字またはサロゲートペアが続きます。書記素および他の Unicode 用語の詳細については、<http://www.unicode.org/>にある『*The Unicode Standard*』を参照してください。

U 形式オペランド用の EXAMINE ステートメントは、通常、コード単位に対して機能します。ただし、CHARPOSITION および CHARLENGTH 節を使用している場合は、書記素シーケンスの開始位置と長さ (コード単位での) を取得できます。返されるコード単位の値は、コード単位のオペランドを必要とする他のステートメントまたは節で使用できます (例: SUBSTRING 節)。

EXAMINE ステートメントのこの構文に関する詳細については、「*Natural* プログラミング言語での Unicode およびコードページのサポート」の「ステートメント」セクションと「EXAMINE」セクションも参照してください。

オペランド定義テーブル:


オペランド	構文要素				フォーマット								オペランド参照	ダイナミック定義								
<i>operand1</i>	C	S	A		U							B								可	不可	
<i>operand2</i>	C	S								N	P	I	B*								可	不可
<i>operand3</i>	C	S								N	P	I	B*								可	不可
<i>operand4</i>	C	S								N	P	I									可	不可
<i>operand5</i>	C	S								N	P	I									可	不可
<i>operand6</i>	C	S								N	P	I									可	不可
<i>operand7</i>	C	S								N	P	I									可	不可

\* *operand2* および *operand3* のフォーマット B は、4 以下の長さでのみ使用できます。

構文要素の説明:

<b>FULL</b>	オペランドに FULL を指定すると、末尾の空白も含めて値全体が処理されます。FULL を指定しない場合は、オペランド内の末尾の空白は無視されます。
<b>SUBSTRING</b>	通常、フィールドの内容はフィールドの始めから終わりまでまたは空白以外の最後の文字まで調べられます。
<i>operand1</i>	
<i>operand2</i>	SUBSTRING オプションを使用すると、フィールドの特定の部分のみを調べることができます。SUBSTRING 節のフィールド名 (オペランド1) の後に、まず開始位置 ( <i>operand2</i> )、次に調べるフィールド部分の長さ ( <i>operand3</i> ) を指定します。 <i>operand2</i> と <i>operand3</i> は、コード単位で指定します。
<i>operand3</i>	

	<p>例えば、フィールド #A の 5 番目から 12 番目を調べるには、次のように指定します。</p> <pre>EXAMINE SUBSTRING (#A,5,8)</pre> <p><b>注意:</b></p> <ol style="list-style-type: none"> <li>1. <i>operand2</i> を省略すると、開始位置は 1 と見なされます。</li> <li>2. <i>operand3</i> を省略すると、長さはフィールドの開始位置から終わりまでと見なされます。</li> <li>3. SUBSTRING を DYNAMIC 変数と一緒に使用すると、フィールドは固定長変数と同様に動作します。つまり、DELETE または REPLACE 操作が実行されたかどうかに関係なく、EXAMINE 操作の結果として長さ (*LENGTH) が変更されることはありません。</li> </ol>
<b>CHARPOSITION</b> <i>operand4</i>	<i>operand4</i> では、書記素シーケンスの開始位置 (Unicode 書記素) を定義します。対応する位置 (コード単位) が、 <i>operand6</i> で返されます。CHARLENGTH 節が指定されている場合は、この節を省略できます。その場合、開始位置は 1 とみなされます。
<b>CHARLENGTH</b> <i>operand5</i>	<i>operand5</i> では、書記素シーケンスの長さ (Unicode 書記素) を定義します。書記素シーケンスの長さ (コード単位) が、 <i>operand7</i> で返されます。CHARPOSITION 節が指定されている場合は、この節を省略できます。その場合は、開始位置から文字列の末尾までの長さが返されます。
<b>GIVING POSITION IN</b> <i>operand6</i>	<i>operand6</i> は、 <i>operand4</i> と <i>operand5</i> によって定義された書記素シーケンスの開始位置 (コード単位) を受け取ります。オペランド 1 が <i>operand4</i> 書記素よりも小さい場合は、0 が返されます。GIVING LENGTH 節が指定されている場合は、この節を省略できます。
<b>GIVING LENGTH IN</b> <i>operand7</i>	<i>operand7</i> は、 <i>operand4</i> と <i>operand5</i> によって定義された書記素シーケンスの長さ (コード単位) を受け取ります。 <i>operand1</i> が <i>operand4+operand5</i> 書記素よりも小さい場合は、0 が返されます。GIVING POSITION 節が指定されている場合は、この節を省略できます。

 **注意:**

1. CHARPOSITION と CHARLENGTH 節の一方または両方を指定する必要があります。
2. GIVING POSITION と GIVING LENGTH 節の一方または両方を指定する必要があります。

## 例

- 例 1 - EXAMINE
- 例 2 - EXAMINE SUBSTRING、PATTERN、TRANSLATE
- 例 3 - EXAMINE TRANSLATE

## ■ 例 4 - Unicode 書記素用の EXAMINE

### 例 1 - EXAMINE

```

** Example 'EXMEX1': EXAMINE
*****
DEFINE DATA LOCAL
1 #TEXT   (A40)
1 #A      (A1)
1 #START  (N2)
1 #NMB1   (N2)
1 #NMB2   (N2)
1 #NMB3   (N2)
1 #NMBEX2 (N2)
1 #NMBEX3 (N2)
1 #NMBEX4 (N2)
1 #POSEX5 (N2)
1 #LGHEX6 (N2)
1 #NMBEX7 (N2)
1 #NMBEX8 (N2)
END-DEFINE
*
WRITE 'EXAMPLE 1 (GIVING NUMBER, WITH DELIMITER)'
MOVE 'ABC  A B C  .A.  .B.  .C.  -A-  -B-  -C-  ' TO #TEXT
ASSIGN #A = 'A'
EXAMINE #TEXT FOR #A GIVING NUMBER #NMB1
EXAMINE #TEXT FOR #A WITH DELIMITER GIVING NUMBER #NMB2
EXAMINE #TEXT FOR #A WITH DELIMITER '.' GIVING NUMBER #NMB3
WRITE NOTITLE '=' #NMB1 '=' #NMB2 '=' #NMB3
*
WRITE / 'EXAMPLE 2 (WITH DELIMITER, REPLACE, GIVING NUMBER)'
WRITE '=' #TEXT
EXAMINE #TEXT FOR #A WITH DELIMITER '-' REPLACE WITH '*'
      GIVING NUMBER #NMBEX2
WRITE '=' #TEXT '=' #NMBEX2
*
WRITE / 'EXAMPLE 3 (REPLACE, GIVING NUMBER)'
WRITE '=' #TEXT
EXAMINE #TEXT ' ' REPLACE WITH '+' GIVING NUMBER #NMBEX3
WRITE '=' #TEXT '=' #NMBEX3
*
WRITE / 'EXAMPLE 4 (FULL, REPLACE, GIVING NUMBER)'
WRITE '=' #TEXT
EXAMINE FULL #TEXT ' ' REPLACE WITH '+' GIVING NUMBER #NMBEX4
WRITE '=' #TEXT '=' #NMBEX4
*
WRITE / 'EXAMPLE 5 (DELETE, GIVING POSITION)'
WRITE '=' #TEXT
EXAMINE #TEXT '+' DELETE GIVING POSITION #POSEX5
WRITE '=' #TEXT '=' #POSEX5
*

```

```

WRITE / 'EXAMPLE 6 (DELETE, GIVING LENGTH)'
WRITE '=' #TEXT
EXAMINE #TEXT FOR 'A' DELETE GIVING LENGTH #LGHEX6
WRITE '=' #TEXT '=' #LGHEX6
*
*
NEWPAGE
*
MOVE 'ABC  A B C  .A.  .B.  .C.  -A-  -B-  -C-  ' TO #TEXT
*
ASSIGN #A = 'A B C'
ASSIGN #START = 6
*
WRITE / 'EXAMPLE 7 (SUBSTRING, GIVING NUMBER)'
WRITE '=' #TEXT
EXAMINE SUBSTRING(#TEXT,#START,9) FOR #A GIVING NUMBER #NMBEX7
WRITE '=' #TEXT '=' #NMBEX7
*
WRITE / 'EXAMPLE 8 (PATTERN, GIVING NUMBER)'
WRITE '=' #TEXT
EXAMINE #TEXT FOR PATTERN '-A-' GIVING NUMBER #NMBEX8
WRITE '=' #TEXT '=' #NMBEX8
*
END

```

プログラム **EXMEX1** の出力：

```

EXAMPLE 1 (GIVING NUMBER, WITH DELIMITER)
#NMB1:  4 #NMB2:  3 #NMB3:  1

EXAMPLE 2 (WITH DELIMITER, REPLACE, GIVING NUMBER)
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B-
#TEXT: ABC  A B C  .A.  .B.  .C.  -*  -B- #NMBEX2:  1

EXAMPLE 3 (REPLACE, GIVING NUMBER)
#TEXT: ABC  A B C  .A.  .B.  .C.  -*  -B-
#TEXT: ABC+++A+B+C+++A.++.B.++.C.++++-*+++B- #NMBEX3:  18

EXAMPLE 4 (FULL, REPLACE, GIVING NUMBER)
#TEXT: ABC+++A+B+C+++A.++.B.++.C.++++-*+++B-
#TEXT: ABC+++A+B+C+++A.++.B.++.C.++++-*+++B-+ #NMBEX4:  1

EXAMPLE 5 (DELETE, GIVING POSITION)
#TEXT: ABC+++A+B+C+++A.++.B.++.C.++++-*+++B-+
#TEXT: ABCABC.A..B..C.-*--B- #POSEX5:  4

EXAMPLE 6 (DELETE, GIVING LENGTH)
#TEXT: ABCABC.A..B..C.-*--B-
#TEXT: BCBC...B..C.-*--B- #LGHEX6:  18

```

EXAMPLE 7 (SUBSTRING, GIVING NUMBER)

```
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B-
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B-  #NMBEX7:  1
```

EXAMPLE 8 (PATTERN, GIVING NUMBER)

```
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B-
#TEXT: ABC  A B C  .A.  .B.  .C.  -A-  -B-  #NMBEX8:  1
```

## 例 2 - EXAMINE SUBSTRING、PATTERN、TRANSLATE

```
** Example 'EXMEX2': EXAMINE TRANSLATE
*****
DEFINE DATA LOCAL
1 #TEXT  (A50)
1 #TAB   (A2/1:10)
1 #START (N2)
END-DEFINE
*
MOVE 'ABC  A B C  .A.  .B.  .C.  -A-  -B-  -C-  ' TO #TEXT
*
MOVE 'AX' TO #TAB(1)
MOVE 'BY' TO #TAB(2)
MOVE 'CZ' TO #TAB(3)
*
*
WRITE 'EXAMPLE 1 (USING TRANSLATION TABLE)'
WRITE '=' #TEXT
EXAMINE #TEXT TRANSLATE USING #TAB(*)
WRITE NOTITLE '=' #TEXT
*
WRITE / 'EXAMPLE 2 (USING INVERTED TRANSLATION TABLE)'
WRITE '=' #TEXT
EXAMINE #TEXT TRANSLATE USING INVERTED #TAB(*)
WRITE NOTITLE '=' #TEXT
*
WRITE / 'EXAMPLE 3 (USING SUBSTRING, LOWER CASE)'
WRITE '=' #TEXT
ASSIGN #START = 13
EXAMINE SUBSTRING(#TEXT,#START,15) TRANSLATE INTO LOWER CASE
WRITE '=' #TEXT
END
```

プログラム EXMEX2 の出力：

```
EXAMPLE 1 (USING TRANSLATION TABLE)
#TEXT: ABC  A B C  .A. .B. .C.  -A- -B- -C-
#TEXT: XYZ  X Y Z  .X. .Y. .Z.  -X- -Y- -Z-

EXAMPLE 2 (USING INVERTED TRANSLATION TABLE)
#TEXT: XYZ  X Y Z  .X. .Y. .Z.  -X- -Y- -Z-
#TEXT: ABC  A B C  .A. .B. .C.  -A- -B- -C-

EXAMPLE 3 (USING SUBSTRING, LOWER CASE)
#TEXT: ABC  A B C  .A. .B. .C.  -A- -B- -C-
#TEXT: ABC  A B C  .a. .b. .c.  -A- -B- -C-
```

### 例 3 - EXAMINE TRANSLATE

```
** Example 'EXMEX2': EXAMINE TRANSLATE
*****
DEFINE DATA LOCAL
1 #TEXT (A50)
1 #TAB (A2/1:10)
1 #START (N2)
END-DEFINE
*
MOVE 'ABC  A B C  .A. .B. .C.  -A- -B- -C- ' TO #TEXT
*
MOVE 'AX' TO #TAB(1)
MOVE 'BY' TO #TAB(2)
MOVE 'CZ' TO #TAB(3)
*
*
WRITE 'EXAMPLE 1 (USING TRANSLATION TABLE)'
WRITE '=' #TEXT
EXAMINE #TEXT TRANSLATE USING #TAB(*)
WRITE NOTITLE '=' #TEXT
*
WRITE / 'EXAMPLE 2 (USING INVERTED TRANSLATION TABLE)'
WRITE '=' #TEXT
EXAMINE #TEXT TRANSLATE USING INVERTED #TAB(*)
WRITE NOTITLE '=' #TEXT
*
WRITE / 'EXAMPLE 3 (USING SUBSTRING, LOWER CASE)'
WRITE '=' #TEXT
ASSIGN #START = 13
EXAMINE SUBSTRING(#TEXT,#START,15) TRANSLATE INTO LOWER CASE
WRITE '=' #TEXT
END
```



プログラム EXMEX2 の出力：

```
EXAMPLE 1 (USING TRANSLATION TABLE)
#TEXT: ABC  A B C  .A. .B. .C.  -A- -B- -C-
#TEXT: XYZ  X Y Z  .X. .Y. .Z.  -X- -Y- -Z-

EXAMPLE 2 (USING INVERTED TRANSLATION TABLE)
#TEXT: XYZ  X Y Z  .X. .Y. .Z.  -X- -Y- -Z-
#TEXT: ABC  A B C  .A. .B. .C.  -A- -B- -C-

EXAMPLE 3 (USING SUBSTRING, LOWER CASE)
#TEXT: ABC  A B C  .A. .B. .C.  -A- -B- -C-
#TEXT: ABC  A B C  .a. .b. .c.  -A- -B- -C-
```

#### 例 4 - Unicode 書記素用の EXAMINE

この例では、文字 ä と ü を含む Unicode 文字列の分析を示します。この 2 つの文字は基本文字として定義され、その後に結合文字が続きます。ä は U+0061 で、その後に U+0308 が続きます。また、ü のコードは U+0075 で、その後に U+0308 が続きます。

```
DEFINE DATA LOCAL
1 #U (U20)
1 #START (I2)
1 #POS (I2)
1 #LEN (I2)
END-DEFINE
#U := U'AB'-UH'00610308'-U'CD'-UH'00750308'-U'EF'
*
REPEAT
  #START := #START + 1
  EXAMINE #U FOR CHARPOSITION #START
          CHARLENGTH 1
          GIVING POSITION IN #POS
          LENGTH IN #LEN
*
  INPUT (AD=0) MARK POSITION #POS IN FIELD *#U
  '          UNICODE-STRING:' #U (AD=MI)
// '          CHARACTER NO.:' #START (EM=9)
/ 'STARTS AT BYTE POSITION:' #POS (EM=9)
/ '          AND THE LENGTH IS:' #LEN (EM=9)
WHILE #POS NE 0
END-REPEAT
END
```

出力：

メインフレーム環境：	Windows、UNIX、およびOpenVMS環境（Natural Web I/O インターフェイスでの出力）：
<pre>                     UNICODE-STRING: ABa?CDu?EF                      CHARACTER NO.: 1 STARTS AT BYTE POSITION: 1                     AND THE LENGTH IS: 1                 </pre>	<pre>                     UNICODE-STRING: ABäCDüEF                      CHARACTER NO.: 1 STARTS AT BYTE POSITION: 1                     AND THE LENGTH IS: 1                 </pre>
<p>Enter キーを押して続行します。</p>	<p>Enter キーを押して続行します。</p>
<pre>                     UNICODE-STRING: ABa?CDu?EF                      CHARACTER NO.: 2 STARTS AT BYTE POSITION: 2                     AND THE LENGTH IS: 1                 </pre>	<pre>                     UNICODE-STRING: ABäCDüEF                      CHARACTER NO.: 2 STARTS AT BYTE POSITION: 2                     AND THE LENGTH IS: 1                 </pre>
<p>Enter キーを押して続行します。</p>	<p>Enter キーを押して続行します。</p>
<p>位置3にある文字は結合文字シーケンスで、コード単位2個分の長さがあることに注意してください。</p>	
<pre>                     UNICODE-STRING: AB&lt;b&gt;a&lt;/b&gt;?CDu?EF                      CHARACTER NO.: 3 STARTS AT BYTE POSITION: 3                     AND THE LENGTH IS: 2                 </pre>	<pre>                     UNICODE-STRING: AB&lt;b&gt;ä&lt;/b&gt;CDüEF                      CHARACTER NO.: 3 STARTS AT BYTE POSITION: 3                     AND THE LENGTH IS: 2                 </pre>
<p>以下、同様。</p>	<p>以下、同様。</p>

# 62 EXPAND

---

▪ 機能 .....	394
▪ 構文説明 .....	394

```
EXPAND { dynamic-clause } [GIVING operand5]  
       { array-clause }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[REDUCE](#) | [RESIZE](#)

関連機能グループ：「[ダイナミック変数または X-array のメモリ管理制御](#)」

## 機能

EXPAND ステートメントは、拡張のために使用されます。

- ダイナミック変数 (*dynamic-clause*) の割り当てられた長さ、または
- X-array (*array-clause*) のオカレンス数

詳細については、『[プログラミングガイド](#)』の次のセクションを参照してください。

[ダイナミック変数の使用](#)

[ダイナミック変数のメモリスペースの割り当て／解放](#)

[X-array](#)

[X-Group 配列のストレージ管理](#)

## 構文説明

オペランド定義テーブル：

オペランド	構文要素		フォーマット										オペランド参照	ダイナミック定義					
<i>operand1</i>	S	A		A	U					B							不可	不可	
<i>operand2</i>	C	S							I								不可	不可	
<i>operand3</i>			A	G	A	U	N	P	I	F	B	D	T	L	C	G	O	可	不可
<i>operand4</i>	C	S					N	P	I									不可	不可
<i>operand5</i>	S								I4									不可	可

構文要素の説明：

<i>dynamic-clause</i>	EXPAND DYNAMIC VARIABLE ステートメントは、ダイナミック変数 ( <i>operand1</i> ) の割り当てられた長さを、指定した長さ ( <i>operand2</i> ) に拡張します。詳細については、後述の「 <a href="#">Dynamic 節</a> 」を参照してください。
<i>operand1</i>	<i>operand1</i> は、サイズを拡張するダイナミック変数です。
<i>operand2</i>	<i>operand2</i> では、拡張するダイナミック変数の長さを指定します。値として、負ではない整数の定数または Integer4 (I4) タイプの変数を指定する必要があります。
<i>array-clause</i>	EXPAND ARRAY ステートメントでは、X-array ( <i>operand3</i> ) のオカレンス数を、( <i>dim</i> [, <i>dim</i> ],[ <i>dim</i> ]) で指定した上下限の範囲内で増加させます。詳細については、以下の「 <a href="#">Array 節</a> 」を参照してください。
<i>operand3</i>	<i>operand3</i> は、オカレンス数が増やされる X-array です。配列のインデックス表記はオプションです。各次元でインデックス表記として使用できるのは、全範囲を示す表記 * のみです。
<i>dim</i> <i>operand4</i>	X-array 拡張の上下限表記 ( <i>operand4</i> またはアスタリスク) は、ここで指定します。現在の上下限の値を使用する必要がある場合は、 <i>operand4</i> の代わりにアスタリスク (*) を指定できます。詳細については、後述の「 <a href="#">次元</a> 」を参照してください。
<b>GIVING</b> <i>operand5</i>	GIVING 節を指定しない場合は、エラー発生時に Natural ランタイムエラー処理がトリガされます。  GIVING 節を指定した場合は、 <i>operand5</i> に、エラー発生時は Natural メッセージ番号、成功時はゼロが含まれます。

## Dynamic 節

```
[SIZE OF] DYNAMIC [VARIABLE] operand1 TO operand2
```

EXPAND DYNAMIC VARIABLE ステートメントでは、ダイナミック変数 (*operand1*) の割り当てられたサイズを、指定した長さ (*operand2*) に拡張します。

*operand2* が、現在割り当てられている *operand1* の長さよりも小さい場合、このダイナミック変数に関してステートメントは無視されます。現在割り当てられているダイナミック変数のサイズ (\*LENGTH) は修正されません。

## Array 節

```
[AND RESET [OCCURRENCES OF] ARRAY operand3 TO (dim[,dim[,dim]])
```

EXPAND ARRAY ステートメントは、X-array (*operand3*) のオカレンス数を、TO(*dim[,dim[,dim]]*) で指定した上下限の範囲内で増加させます。

RESET オプションは、サイズ変更した X-array のすべてのオカレンスをデフォルトのゼロ値にリセットします。デフォルト (RESET オプションなし) では、実際の値は保持され、サイズ変更した (新しい) オカレンスがリセットされます。

EXPAND ステートメントを使用する場合は、オカレンス数の増加だけが可能です。要求した数が現在割り当てられているオカレンス数よりも小さい場合は、無視されます。

EXPAND ステートメントで使用される上限または下限は、配列に定義された対応する上限または下限と正確に一致している必要があります。

例：

```
DEFINE DATA LOCAL
1 #a(I4/1:*)
1 #g(1:*)
  2 #ga(I4/1:*)

1 #i(i4)
END-DEFINE
...
/* allocating #a(1:10)
EXPAND ARRAY #a TO (1:10)          /* #a is allocated 10
EXPAND ARRAY #a TO (*:10)         /* occurrences.

/* allocating #ga(1:10,1:20)
EXPAND ARRAY #g TO (1:10)         /* 1st dimension is set to (1:10)
EXPAND ARRAY #ga TO (*:*,1:20)   /* 1st dimension is dependent and
                                  /* therefore kept with (*:*)
                                  /* 2nd dimension is set to (1:20)

EXPAND ARRAY #a TO (5:10)         /* This is rejected because the lower index
                                  /* must be 1 or *
EXPAND ARRAY #a TO (#i:10)       /* This is rejected because the lower index
                                  /* must be 1 or *

EXPAND ARRAY #ga TO (1:10,1:20)  /* (1:10) for the 1st dimension is rejected
                                  /* because the dimension is dependent and
                                  /* must be specified with (*:*)
```

詳細については、次を参照してください。

- X-array のストレージ管理
- X-Group 配列のストレージ管理

## 次元

**Array** 節で指定する各次元 (*dim*) は、次の構文を使用して定義します。

$$\left\{ \begin{array}{l} \text{operand4} \\ * \end{array} \right\} \left\{ \begin{array}{l} \text{operand4} \\ * \end{array} \right\}$$

X-array 拡張の上下限表記 (*operand4* またはアスタリスク) は、ここで指定します。現在の上下限の値を使用する必要がある場合は、*operand4* の代わりにアスタリスク (\*) を指定できます。また、\*:\* の代わりに、1つのアスタリスクを指定することもできます。

次元数 (*dim*) は、X-array の定義された次元数 (1、2、または3) と正確に一致している必要があります。

指定した次元のオカレンス数が、現在割り当てられているオカレンス数より小さい場合は、対応する次元のオカレンス数は変更されません。





# 63      FETCH

---

▪ 機能 .....	400
▪ 構文説明 .....	401
▪ 例 .....	402

```
FETCH [ { REPEAT } | { RETURN } ] operand1 [operand2 [(parameter)]] ...
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CALL](#) | [CALL FILE](#) | [CALL LOOP](#) | [CALLNAT](#) | [DEFINE SUBROUTINE](#)  
| [ESCAPE](#) | [FETCH](#) | [PERFORM](#)

関連機能グループ：「[プログラムおよびルーチンの呼び出し](#)」

## 機能

FETCH ステートメントは、メインプログラムとして書かれた Natural オブジェクトプログラムを実行します。ロードするプログラムは、あらかじめ、Natural システムファイルに CATALOG あるいは STOW コマンドで保存しておく必要があります。FETCH ステートメントの実行によって、Natural ソースワークエリア内にあるソースプログラムが置き換えられることはありません。

Natural RPC については、『[Natural リモートプロシージャコール \(RPC\)](#)』ドキュメントの「[サーバーに対する Natural ステートメントの注意事項](#)」を参照してください。

### その他の考慮事項

FETCH 先プログラムは、FETCH で渡されたパラメータの他に、グローバルデータエリアにもアクセスできます。

FETCH ステートメントは、Natural 管理者が設定した Natural プロファイルパラメータ OPRB に基づいて、内部的に [END TRANSACTION](#) ステートメントを実行する場合があります。そのため、1 論理トランザクションが複数の Natural プログラムにおよぶ場合は、OPRB パラメータが正しく設定されているかどうかを確認する必要があります。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A	可	不可
<i>operand2</i>	C S A G	A U N P I F B D T L G	可	可

構文要素の説明：

<b>REPEAT</b>	<p>ユーザー操作の必要性の排除：</p> <p>REPEAT 節は、FETCH されたプログラムの実行中に発行される各 <b>INPUT</b> ステートメントに対して、ユーザーの入力待ちを省略します。これにより、ユーザーが Enter キーを押すことなく、プログラムの実行についての情報を端末に送ることができます。</p>
<b>RETURN</b>	<p>プログラムタイプのオブジェクトのルーチンとして呼び出しおよび実行：</p> <p>RETURN 指定がないと、FETCH ステートメントを発行したプログラムの実行を終了し、FETCH 先のプログラムが「メインプログラム」（レベル 1）として処理されます。</p> <p>RETURN を指定すると、FETCH を発行したプログラムの実行は中断され（終了はしない）、FETCH 先のプログラムは「次のレベルのプログラム」として処理が始まります。そして、FETCH 先のプログラムで <b>END</b> ステートメントや <b>ESCAPE ROUTINE</b> ステートメントが現れると、FETCH RETURN ステートメントを発行したプログラムに制御が戻されます。処理は FETCH RETURN ステートメントの後のステートメントに渡ります。</p>
<i>operand1</i>	<p>プログラム名：</p> <p>プログラムモジュール名（最大 8 文字）は、英数字定数で指定するか、英数字フォーマットのユーザー定義変数（長さ 1~8）の内容として指定します。</p> <p>Natural は、FETCH が発行されたときに稼働しているライブラリ内でプログラムを検索します。プログラムが見つからないとき、Natural は STEPLIB を検索します。それでも見つからないときには、エラーメッセージを発行します。</p> <p>プログラム名にアンパサンド (&amp;) を含むこともできます。これは、実行時にシステム変数 *LANGUAGE の現在の値で置き換えられます。これにより、入力時の言語に応じて入力処理ごとに異なるプログラムを呼び出すことができるようになります。</p>
<i>operand2</i>	<p>パラメータフィールドの引き渡し：</p> <p>FETCH ステートメントで呼び出されるプログラムに、パラメータフィールドを渡すこともできます。パラメータフィールドはどのフォーマットでも定義できます。パラメータは、対応する INPUT フィールドに適したフォーマットに変換されます。すべてのパラメータは、Natural スタックの最上位に配置されます。</p>

	<p>FETCH 先のプログラムは、<b>INPUT</b> ステートメントでパラメータフィールドを読み込みます。FETCH 先のプログラムの最初の INPUT ステートメントで指定したフィールドに、全パラメータフィールド値が入ります。数字フォーマットで定義されたパラメータフィールドに対しては、INPUT ステートメント側で SG=ON の指定が必要です。それは FETCH ステートメントで数字フォーマットで定義された各パラメータフィールドは、値が負の場合は符号桁を持つからです。</p> <p>次の <b>INPUT</b> ステートメントで読み取られる以上のパラメータが渡されると、余分のパラメータは無視されます。パラメータの個数は、Natural システム変数 *DATA で取得できます。</p> <p><b>注意:</b> <i>operand2</i> が時刻変数（フォーマット T）の場合は、変数内容のうち時刻コンポーネントのみが渡され、日付コンポーネントは渡されません。</p>
<i>parameter</i>	<p>日付変数の日付形式：</p> <p><i>operand2</i> が日付変数の場合は、この変数に対する <i>parameter</i> として、セッションパラメータ DF を指定できます。</p>

## 例

呼び出し元のプログラム：

```

** Example 'FETEX1': FETCH (with parameter)
*****
DEFINE DATA LOCAL
1 #PNUM (N8)
1 #FNC (A1)
END-DEFINE
*
INPUT 10X 'SELECTION MENU FOR EMPLOYEES SYSTEM' /
      10X '-' (35) //
      10X 'ADD      (A)' /
      10X 'UPDATE  (U)' /
      10X 'DELETE  (D)' /
      10X 'STOP    (.)' //
      10X 'PLEASE ENTER FUNCTION: ' #FNC ///
      10X 'PERSONNEL NUMBER:' #PNUM
*
DECIDE ON EVERY VALUE OF #FNC
  VALUE 'A', 'U', 'D'
    IF #PNUM = 0
      REINPUT 'PLEASE ENTER A VALID NUMBER' MARK *#PNUM
    END-IF
  VALUE 'A'
    FETCH 'FETEXAD' #PNUM
  VALUE 'U'
    FETCH 'FETEXUP' #PNUM
  VALUE 'D'
    FETCH 'FETEXDE' #PNUM

```

```

VALUE '.'
  STOP
  NONE
  REINPUT 'PLEASE ENTER A VALID FUNCTION' MARK *#FNC
END-DECIDE
*
END

```

#### 呼び出し先のプログラム **FETEXAD** :

```

** Example 'FETEXAD': FETCH (called by FETEX1)
*****
DEFINE DATA LOCAL
1 #PERS-NR (N8)
END-DEFINE
*
INPUT #PERS-NR
*
WRITE *PROGRAM 'Record added with personnel number:' #PERS-NR
*
END

```

#### 呼び出し先のプログラム **FETEXUP** :

```

** Example 'FETEXUP': FETCH (called by FETEX1)
*****
DEFINE DATA LOCAL
1 #PERS-NR (N8)
END-DEFINE
*
INPUT #PERS-NR
*
WRITE *PROGRAM 'Record updated with personnel number:' #PERS-NR
*
END

```

#### 呼び出し先のプログラム **FETEXDE** :

```

** Example 'FETEXDE': FETCH (called by FETEX1)
*****
DEFINE DATA LOCAL
1 #PERS-NR (N8)
END-DEFINE
*
INPUT #PERS-NR
*
WRITE *PROGRAM 'Record deleted with personnel number:' #PERS-NR
*
END

```

プログラム **FETEX1** の出力：

```
SELECTION MENU FOR EMPLOYEES SYSTEM
```

```
-----  
ADD      (A)  
UPDATE   (U)  
DELETE   (D)  
STOP     (.)
```

```
PLEASE ENTER FUNCTION: D
```

```
PERSONNEL NUMBER: 1150304
```

機能と社員番号を入力した後：

```
Page      1
```

```
05-01-13 11:58:46
```

```
FETEXDE Record deleted with personnel number: 1150304
```

# 64 FIND

---

▪ 機能 .....	406
▪ 制限 .....	408
▪ 構文説明 .....	408
▪ 例 .....	430

FIND	<pre>       {         ALL         (operand1)         FIRST         NUMBER         UNIQUE       } </pre>	<pre> [<i>MULTI-FETCH-clause</i>][RECORDS][IN] [FILE] <i>view-name</i> </pre>
	<pre> [PASSWORD=<i>operand2</i>] [CIPHER=<i>operand3</i>] [WITH] [[LIMIT] (<i>operand4</i>)] <i>basic-search-criterion</i> [COUPLED-clause] ... 4/42 [STARTING WITH ISN=<i>operand5</i>] [<i>SORTED-BY-clause</i>] [<i>RETAIN-clause</i>] [<i>WHERE-clause</i>] [<i>IF-NO-RECORDS-FOUND-clause</i>] <i>statement ...</i> </pre>	
END-FIND		(ストラクチャードモードのみ)
[LOOP]		(レポートモードのみ)

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

FIND ステートメントは、ディスクリプタ（キー）として定義したフィールドの検索条件をもとに、データベースからレコードの集合を選択します。

このステートメントは処理ループを開始させ、選択された各レコードに対してループ内の処理が行われます。各レコードのすべてのフィールドを処理ループ内で参照できます。そのため、選択した各レコード内のフィールドを参照するために、FIND ステートメントに続けて [READ](#) ステートメントを発行する必要はありません。

『プログラミングガイド』の「[FIND ステートメント](#)」も参照してください。



## データベース固有の考慮事項

<b>DL/I</b>	与えられたセグメントオカレンスの、最後のバイトの次から始まるフィールドをアクセスするとき、このフィールドのストレージコピーは、そのフォーマット（数字、空白など）に合わせて埋められます。FIND ステートメントに関する記述中のレコードという用語を、セグメントオカレンスという用語に置き換える必要があります。
<b>VSAM</b>	FIND ステートメントは、キー順（KSDS）およびエントリ順（ESDS）の VSAM データセットにのみ有効です。ESDS データセットには、ベースクラスタの代替インデックスを定義する必要があります。
<b>SQL</b>	<p>FIND FIRST は使用できません。PASSWORD、CIPHER、COUPLED、RETAIN の各節も使用できません。</p> <p>FIND UNIQUE は許可されていません。（例外：FIND UNIQUE はプライマリキーに対して使用できますが、これは互換性のためだけであるため使用しないでください）。</p> <p>SORTED BY 節は SQL の ORDER BY 節に相当します。</p> <p>SQL データベーステーブルに対する基本検索条件は Adabas ファイルの場合と同じ方法で指定できます。本文で使用する用語レコードは SQL 用語の「行」に相当します。</p>

## FIND ステートメントで使用可能な Natural システム変数

発行された FIND ステートメントごとに \*ISN、\*NUMBER および \*COUNTER の各 Natural システム変数が自動的に生成されます。実行中の処理ループの外側や、FIND UNIQUE、FIND FIRST、FIND NUMBER のステートメントで生成されたシステム変数を参照する場合は、参照番号を指定する必要があります。これらのシステム変数のフォーマットおよび長さは P10 です。これを変更することはできません。

<b>*ISN</b>	Adabas	現在処理中のレコードの Adabas 内部シーケンス番号 (ISN) が *ISN に含まれます。FIND NUMBER ステートメントでは、*ISN は使用できません。
	VSAM	『システム変数』ドキュメントの「VSAM に対する *ISN」を参照してください。
	DL/I および SQL	*ISN は使用できません。
	Entire System Server	*ISN は使用できません。
<b>*NUMBER</b>	Adabas	システム変数 *NUMBER には、WITH 節で指定された基本検索条件を満たすレコード件数が含まれます。
	VSAM	『システム変数』ドキュメントの「VSAM に対する *NUMBER」を参照してください。
	DL/I	『システム変数』ドキュメントの「DL/I に対する *NUMBER」を参照してください。
	Entire System Server	*NUMBER は使用できません。
<b>*COUNTER</b>	システム変数 *COUNTER には、入力した処理ループの回数が含まれます。	

「例 13 - *FIND* ステートメントで使用可能な *Natural* システム変数」も参照してください。

### 複数の FIND ステートメント

複数の FIND ステートメントを発行してループのネスト構造を作成することができます。その際に、外側の選択レコードごとに内側のループ処理が行われます。

「例 14 - 複数の *FIND* ステートメント」も参照してください。

## 制限

Entire System Server では、*FIND NUMBER* および *FIND UNIQUE* は使用できません。PASSWORD、CIPHER、COUPLED、RETAIN の各節も使用できません。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	N P I B*	可	不可
<i>operand2</i>	C S	A	可	不可
<i>operand3</i>	C S	N	可	不可
<i>operand4</i>	C S	N P I B*	可	不可
<i>operand5</i>	C S	N P I B*	可	不可

\* *operand1*、*operand4* および *operand5* のフォーマット B は、4 以下の長さでのみ使用できません。

構文要素の説明：

<b>ALL</b> <i>operand1</i>	<p>処理制限：</p> <p>選択された集合から処理するレコード件数は、<i>operand1</i> として定数（0～4294967295 の範囲）や数値のユーザー定義変数名をカッコで囲んで指定することで制限できます。</p> <p>ALL は、オプションで指定してもよく、選択されたレコードをすべて処理することを強調するものです。</p>
----------------------------	--

	<p><i>operand1</i>によって制限が指定された場合、この制限は FIND で開始した処理ループに適用されます。WHERE 節で除かれたレコードは、この制限値の対象とはなりません。</p> <pre>FIND (5) IN EMPLOYEES WITH ...  MOVE 10 TO #CNT(N2) FIND (#CNT) EMPLOYEES WITH ...</pre> <p>このステートメントでは、指定した制限が、LIMIT ステートメントで設定した制限より優先されます。</p> <p>LT パラメータで設定された制限の方が小さい場合は、LT 制限が有効になります。</p> <p><b>注意:</b></p> <ol style="list-style-type: none"> <li>4 桁のレコード件数を指定する場合は、リーディングゼロを指定してください (0nnnn)。それは、カッコで囲まれた 4 桁の数はすべて、ステートメントを参照する行番号であると Natural が解釈するためです。</li> <li><i>operand1</i> は RETAIN 節で保持される ISN 集合のサイズには影響しません。<i>operand1</i> は FIND ループに入るときに評価されます。<i>operand1</i> の値が FIND ループ内で変更されても、この値は処理するレコード数に影響を与えません。</li> </ol>
<p><b>FIND FIRST   FIND NUMBER   FIND UNIQUE</b></p>	<p>これらのオプションは、</p> <ul style="list-style-type: none"> <li>■ 選択されたレコードの集合の最初の1件を使用する (<b>FIND FIRST</b>)、</li> <li>■ 選択された集合のレコード件数を得る (<b>FIND NUMBER</b>)、</li> <li>■ または 1 レコードだけが検索条件を満たすことを保証する (<b>FIND UNIQUE</b>) ために使用します。</li> </ul> <p>これらのオプションの詳細な説明については、下記を参照してください。</p>
<p><i>MULTI-FETCH-clause</i></p>	<p>Adabas データベースの場合、Natural はデータベースアクセス当たり 1 回以上の読み取りを許可する MULTI-FETCH 節を提供します。詳細については、「<b>MULTI-FETCH 節</b>」を参照してください。</p>
<p><i>view-name</i></p>	<p>DEFINE DATA ステートメント内、またはグローバルデータエリアやローカルデータエリアで定義されたビュー名です。</p> <p>レポートモードでは、<i>view-name</i> を DDM の名前にすることもできます。</p>
<p><b>PASSWORD=<i>operand2</i></b></p>	<p><b>PASSWORD 節:</b></p> <p>PASSWORD 節は、Adabas または VSAM データベースにのみ適用されます。この節は Entire System Server では使用できません。</p>

	<p>PASSWORD 節は、パスワード保護された Adabas または VSAM ファイルのデータを読み込むときまたは書き込むときに、そのパスワード (<i>operand2</i>) を指定するために使用します。パスワードで保護されたファイルにアクセスするときは、データベースセキュリティ責任者にパスワードの使用や割り当てについて相談してください。</p> <p>パスワードを定数で指定する場合、PASSWORD 節は必ずソースコード行の第 1 桁目からコーディングします。また、キーワード PASSWORD と等号の間には空白を入れないでください。そうすれば、プログラムのソースコードをリストしたときに、サイファキーはリストに出力されません。</p> <p>TP モードでは、PASSWORD 節を打ち込む前に %* 端末コマンドを入力すれば、パスワードは表示されません。</p> <p>PASSWORD 節を省略すると、<b>PASSW</b> ステートメントで指定したパスワードが適用されます。</p> <p>パスワードの値は、処理ループの実行中に変更しないでください。</p> <p>「<a href="#">例 1 - PASSWORD 節</a>」も参照してください。</p>
<p><b>CIPHER=</b><i>operand3</i></p>	<p><b>CIPHER 節：</b></p> <p>CIPHER 節は Adabas データベースにのみ適用されます。この節は Entire System Server では使用できません。</p> <p>CIPHER 節は、サイファリング (暗号化) された Adabas ファイルからデータを読み込むとき、そのサイファキー (<i>operand3</i>) を指定します。サイファリングされたファイルにアクセスするときは、データベースセキュリティの責任者にサイファキーの使用や割り当てについて相談してください。</p> <p>サイファキーは、数値定数 (8 桁) またはユーザー定義変数 (N8) で指定できます。</p> <p>サイファキーを定数で指定する場合、CIPHER 節は必ずソースコード行の第 1 桁目からコーディングします。そうすれば、プログラムのソースコードをリストしたときに、サイファキーはリストに出力されません。TP モードでは、CIPHER 節を打ち込む前に %* 端末コマンドを入力すると、CIPHER 節は表示されません。</p> <p>サイファキーの値は、FIND ステートメントで開始された処理ループ内では変更しないでください。</p> <p>「<a href="#">例 2 - CIPHER 節</a>」も参照してください。</p>
<p><b>WITH LIMIT</b> <i>operand4</i> <i>basic-search-criterion</i></p>	<p><b>WITH 節：</b></p> <p>WITH 節は必須です。データベースで定義されているキーフィールド (ディスクリプタ) から構成される <i>basic-search-criterion</i> を指定するために使用されます (「<a href="#">Adabas ファイルに対する基本検索条件</a>」を参照)。</p>

	次のデータベース固有の考慮事項が適用されます。						
	<table border="1"> <tr> <td>Adabas ファイルの場合：</td> <td>WITH節でディスクリプタ、サブディスクリプタ、スーパーディスクリプタ、ハイパーディスクリプタ、およびフォネティックディスクリプタを使用できます。非ディスクリプタ（DDMにNと設定されたフィールド）も指定できます。</td> </tr> <tr> <td>DL/I ファイルの場合：</td> <td>DDMでDと設定されたキーフィールドだけを検索条件に使用できます。</td> </tr> <tr> <td>VSAM ファイルの場合：</td> <td>VSAMのキーフィールドだけを使用できます。</td> </tr> </table> <p>WITH節で選択するレコード件数は、キーワードLIMITとともに数値定数やユーザー定義変数をカッコで囲んだ制限値（operand4）で指定できます。選択レコード件数が制限値を超過した場合、エラーメッセージが出力されプログラムが終了します。</p> <p><b>注意:</b> 制限が4桁の数の場合、リーディングゼロを指定する必要があります（0nnnn）。これは、カッコで囲まれた4桁の数は、すべてステートメントを参照する行番号であるとNaturalが解釈するためです。</p>	Adabas ファイルの場合：	WITH節でディスクリプタ、サブディスクリプタ、スーパーディスクリプタ、ハイパーディスクリプタ、およびフォネティックディスクリプタを使用できます。非ディスクリプタ（DDMにNと設定されたフィールド）も指定できます。	DL/I ファイルの場合：	DDMでDと設定されたキーフィールドだけを検索条件に使用できます。	VSAM ファイルの場合：	VSAMのキーフィールドだけを使用できます。
Adabas ファイルの場合：	WITH節でディスクリプタ、サブディスクリプタ、スーパーディスクリプタ、ハイパーディスクリプタ、およびフォネティックディスクリプタを使用できます。非ディスクリプタ（DDMにNと設定されたフィールド）も指定できます。						
DL/I ファイルの場合：	DDMでDと設定されたキーフィールドだけを検索条件に使用できます。						
VSAM ファイルの場合：	VSAMのキーフィールドだけを使用できます。						
<i>COUPLED-clause</i>	この節は、Adabasのカップリング機能を使用した検索を行います。「 <b>COUPLED節</b> 」を参照してください。						
<b>STARTING WITH ISN=operand5</b>	この節は、処理を継続する次のレコードを簡単に指定するために、中断されたFINDループ内での再位置決めを使用できます。「 <b>STARTING WITH節</b> 」を参照してください。						
<i>SORTED-BY-clause</i>	<b>SORTED BY</b> 節は、選択したレコードを1個～3個のディスクリプタを基準にAdabasソートします。「 <b>SORTED BY節</b> 」を参照してください。						
<i>RETAIN-clause</i>	この節を使用して、大規模ファイルの検索結果を以降の処理のために保持しておくことができます。「 <b>RETAIN節</b> 」を参照してください。						
<i>WHERE-clause</i>	この節を使用して、追加の選択条件（logical-condition）を指定できます。「 <b>WHERE節</b> 」を参照してください。						
<i>IF-NO-RECORDS-FOUND-clause</i>	<b>IF NO RECORDS FOUND</b> 節を使用すると、 <b>WITH</b> 節および <b>WHERE</b> 節で指定された選択条件を満たすレコードがない場合でも、FINDステートメントで始まる処理ループに入ります。「 <b>IFNO RECORDS FOUND節</b> 」を参照してください。						
<b>END-FIND</b>	FINDステートメントを終了するには、Naturalの予約キーワード <b>END-FIND</b> を使用する必要があります。						

## FIND FIRST

FIND FIRST ステートメントは、WITH 条件および WHERE 条件を満たす最初の 1 件のレコードを選択し、処理します。

Adabas データベースでは、選択レコードのうち最も低い Adabas ISN 値を持つレコードが処理されます。

このステートメントは処理ループを開始しません。

### 制限

- FIND FIRST はレポーティングモードでのみ使用可能です。
- FIND FIRST は、DL/I および SQL データベースでは使用できません。
- IF NO RECORDS FOUND 節は FIND FIRST ステートメントでは使用できません。

### FIND FIRST で使用可能なシステム変数

次の Natural システム変数が FIND FIRST ステートメントで使用できます。

*ISN	システム変数 *ISN には選択レコードの Adabas ISN が含まれます。WITH および WHERE 条件の評価後にレコードが 1 件も見つからなかった場合、*ISN はゼロになります。  VSAM データベースあるいは Entire System Server では、*ISN を使用できません。
*NUMBER	システム変数 *NUMBER には、WITH 条件の評価後、WHERE 条件が評価される前のレコード数が含まれます。WITH 条件を満たすレコードがなければ、*NUMBER はゼロになります。  Entire System Server では *NUMBER を使用できません。
*COUNTER	レコードが見つかった場合は 1、レコードが見つからなかった場合は 0 が、システム変数 *COUNTER に含まれます。

FIND FIRST ステートメントの例：[FNDFIR](#) プログラム（レポーティングモード）を参照してください。

## FIND NUMBER

FIND NUMBER ステートメントは、指定した WITH/WHERE 条件を満たすレコード件数を得るために使用します。処理ループは開始されず、データベースのフィールドも使用できません。



**注意:** WHERE 節を使用すると、負荷が生じることがあります。

### 制限

- FIND NUMBER ステートメントに、[SORTED BY](#) や [IF NO RECORDS FOUND](#) の各節を指定することはできません。
- ストラクチャードモードでは、[WHERE](#) 節を使用できません。

- FIND NUMBER は、DL/I データベースまたは Entire System Server では使用できません。

## FIND NUMBER で使用可能なシステム変数

次の Natural システム変数が FIND NUMBER ステートメントで使用できます。

*NUMBER	システム変数 *NUMBER には、WITH 条件の評価後に見つかったレコード数が含まれます。
*COUNTER	システム変数 *COUNTER には、WHERE 条件の評価後に見つかったレコード数が含まれます。 *COUNTER は、FIND NUMBER ステートメントに WHERE 節が指定された場合にのみ有効です。

FIND NUMBER の例：[FNDNUM](#) プログラム（レポートモード）を参照してください。

## FIND UNIQUE

FIND UNIQUE ステートメントは、処理のために選択されたレコードが1件であることを保証するために使用します。処理ループは開始しません。WHERE 節の指定がある場合、WHERE 節を評価する内部処理ループが自動的に生成されます。

条件を満たすレコードがないとき、あるいは条件を満たすレコードが複数あるときは、エラーメッセージが返されます。ON ERROR ステートメントで、エラー条件をチェックできます。

## FIND UNIQUE で使用できるシステム変数

*ISN	システム変数 *ISN は、レコードの一意な ISN 番号を持ちます。レコード自体が一意である必要があります。
*NUMBER	システム変数 *NUMBER は、有効な FIND UNIQUE 実行に対しては常に 1 です。 エラーが発生した場合、*NUMBER は、他の値 (=0 または >=2) を含んでいる可能性があります。このエラー条件は、ON ERROR ステートメントで使用することができます。WHERE 節が指定されていない場合、*NUMBER は評価されません。
*COUNTER	システム変数 *COUNTER には、WHERE 条件の評価後に見つかったレコード数が含まれます。WHERE 節が指定されていない場合、*COUNTER は評価されません。

## 制限事項

- FIND UNIQUE はレポートモードでのみ使用可能です。
- FIND UNIQUE は、DL/I データベースまたは Entire System Server では使用できません。
- SQL データベースに、FIND UNIQUE は使用できません（例外：メインフレーム環境では、FIND UNIQUE はプライマリキーに対して使用できますが、これは互換性のためだけであるため使用しないでください）。
- SORTED BY や IF NO RECORDS FOUND の各節を FIND UNIQUE ステートメントで使用しないでください。



FIND UNIQUE の例：[FNDUNQ](#) プログラム（レポートモード）を参照してください

### MULTI-FETCH 節



**注意:** この節は、Adabas または DB2 データベースでのみ使用できます。

```
[ MULTI-FETCH { ON
                { OFF
                { OF multi-fetch-factor } } ]
```

詳細については、『プログラミングガイド』の「MULTI-FETCH 節」（Adabas）または『データベース管理システムインターフェイス』ドキュメントの「Natural for DB2」で「複数行の処理」（SQL）を参照してください。

### Adabas ファイルに対する基本検索条件

1 <i>descriptor</i> [[ <i>i</i> ]]	EQ = EQUAL EQUAL TO	<i>value</i>	{ OR { EQ = EQUAL EQUAL TO } ... } THRU <i>value</i> [BUT NOT <i>value</i> [THRU <i>value</i> ]]
2 <i>descriptor</i> [[ <i>i</i> ]]	EQ = EQUAL EQUAL TO NE ≠ < > NOT = NOT EQ NOTEQUAL NOT EQUAL NOT EQUAL TO LT LESS THAN < GE GREATER EQUAL >= NOT < NOT LT GREATER THAN > LE LESS EQUAL <=	<i>value</i>	



NOT > NOT GT
3 <i>set-name</i>

オペランド定義テーブル：

オペランド	構文要素	フォーマット											ステートメント参照	ダイナミック定義	
<i>descriptor</i>	S A	A	N	P	I	F	B	D	T	L				不可	不可
<i>value</i>	C S	A	N	P	I	F	B	D	T	L				可	不可
<i>set-name</i>	C S	A												不可	不可


構文要素の説明：

<i>descriptor</i>	Adabas ディスクリプタ、サブディスクリプタ、スーパーディスクリプタ、ハイパーディスクリプタ、またはフォネティックディスクリプタ。DDMでディスクリプタ以外としてマークされたフィールドも指定できます。
(i)	ピリオディックグループに含まれるディスクリプタをインデックス付きまたはなしで指定できます。インデックスなしで指定した場合、レコードは指定した値がいずれかのオカレンスに見つかった場合に選択されます。インデックス付きで指定した場合、レコードは値がインデックスで指定したオカレンス内に見つかった場合にのみ選択されます。インデックスは定数で指定する必要があります。インデックス範囲は使用できません。  マルチプルバリューフィールドのディスクリプタは、インデックスなしで指定する必要があります。レコードは、値がレコード内で位置に関係なく見つかった場合に選択されます。
<i>value</i>	検索条件値。ディスクリプタと検索値のフォーマットは変換可能にする必要があります。
<i>set-name</i>	RETAIN 節を指定した FIND ステートメントで、すでに選択されたレコードの集合の名前です。FIND で参照する集合名は、同じ物理 Adabas ファイルから作成する必要があります。 <i>set-name</i> は、最大 32 文字の文字定数、あるいは英数字フォーマットの変数として指定します。 <i>set-name</i> は、Entire System Server では使用できません。

以下の項目も参照してください。

- **例 3 - WITH 節の基本検索条件**
- **例 4 - マルチプルバリューフィールドの基本検索条件**

NULL インジケータでの検索条件



オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>null-indicator</i>	S	I	不可	不可
<i>value</i>	C S	N P I F B	可	不可

構文要素の説明：


<i>null-indicator</i>	空値インジケータ。	
<i>value</i>	指定可能な値	
	-1	対応するフィールドは値を含んでいません。
	0	対応するフィールドは値を含んでいます。

検索条件の結合 (Adabas ファイル使用時)

*Basic-search-criteria* は、ブール演算子 AND、OR および NOT を使用して結合できます。カッコで囲んで評価の優先順位を制御することもできます。評価の順序は次のとおりです。

1. ()：カッコ
2. NOT：否定 ([2] の形式の *basic-search-criterion* にのみ可)
3. AND：AND 結合
4. OR：OR 結合

*Basic-search-criteria* は論理演算子で複合 *search-expression* の形に結合できます。複合 *search-expression* の構文は次のようになります。



「例5 - WITH 節のさまざまな複合検索式の例」も参照してください。

## ディスクリプタ（キー）の使用

Adabas ユーザーは、ディスクリプタとして定義されているデータベースフィールドを基本検索条件に使用できます。

### サブディスクリプタ、スーパーディスクリプタ、ハイパーディスクリプタ、フォネティックディスクリプタ

Adabas では、検索条件にサブディスクリプタ、スーパーディスクリプタ、ハイパーディスクリプタ、およびフォネティックディスクリプタが使用できます。

- サブディスクリプタは、フィールドの一部で作られたディスクリプタです。
- スーパーディスクリプタは、1つ以上のフィールドまたはフィールドの一部から構成されたディスクリプタです。
- ハイパーディスクリプタは、ユーザー定義アルゴリズムを使用して作ったディスクリプタです。
- フォネティックディスクリプタは、発音による検索（例：人の名前）を行うディスクリプタです。フォネティック検索では、発音が検索値に類似するすべての値が返されます。

どのファイルのどのフィールドをディスクリプタ、サブディスクリプタ、スーパーディスクリプタ、ハイパーディスクリプタ、フォネティックディスクリプタとして使用できるかは対応するDDMに定義されます。

### サブディスクリプタ、スーパーディスクリプタ、フォネティックディスクリプタに対する値

これらのディスクリプタ値は、ディスクリプタの内部フォーマットに対して変換できる必要があります。サブディスクリプタの内部フォーマットは、そのサブディスクリプタを作り出したもとのフィールドのフォーマットと同じです。スーパーディスクリプタの内部フォーマットは、もとのフィールドがすべて数字フォーマットのときはバイナリであり、そうでなければ英数字フォーマットです。フォネティックディスクリプタは、常に英数字フォーマットです。

サブディスクリプタやスーパーディスクリプタの値は、次のように指定できます。

- 数字または16進定数。ただしバイナリフォーマットのスーパーディスクリプタに対しては、16進定数を使用する必要があります（上記参照）。
- サブディスクリプタやスーパーディスクリプタの各部分を、REDEFINE ステートメントで再定義した形のユーザー定義変数内の値。

### データベース配列内のディスクリプタの使用


データベース内の配列に含まれるディスクリプタも、基本検索条件の中に使用できます。Adabasデータベースを使用している場合、MUまたはPEに含まれるディスクリプタのことです。

ピリオディックグループに含まれるディスクリプタをインデックス付きまたはなしで指定できます。インデックスなしで指定した場合、レコードは指定した値がいずれかのオカレンスに見つかった場合を選択されます。インデックス付きで指定した場合、レコードは値がインデックスで指定したオカレンス内に見つかった場合にのみ選択されます。インデックスは定数で指定する必要があります。インデックス範囲は使用できません。

マルチプルバリューフィールドのディスクリプタは、インデックスなしで指定する必要があります。レコードは、値がレコード内で位置に関係なく見つかった場合を選択されます。

「例6 - データベース配列のさまざまな使用例」も参照してください。

### VSAM ファイルに対する基本検索条件



$2 \text{ descriptor} \left\{ \begin{array}{l} \text{EQ} \\ = \\ \text{EQUAL} \\ \text{EQUAL TO} \end{array} \right\} \text{value THRUvalue}$
---

オペランド定義テーブル：


オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>descriptor</i>	S A	A N P B	不可	不可
<i>value</i>	C S	A N P B	可	不可

構文要素の説明：

<i>descriptor</i>	ディスクリプタは、VSAMキーフィールドとしてVSAMファイルに定義され、DDM内にP（プライマリキー）またはA（オルタネートキー）を設定する必要があります。
<i>value</i>	検索値。

*descriptor* と検索 *value* のフォーマットは変換可能にする必要があります。

### DL/I ファイルに対する基本検索条件



オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>descriptor</i>	S A	A N P B	不可	不可
<i>value</i>	C S	A N P B	可	不可

構文要素の説明：

<i>descriptor</i>	ディスクリプタは、DL/I 内で定義されたフィールドで、DDM 内に D と設定されている必要があります。
<i>value</i>	検索値。

HDAM データベースに対しては、次の *basic-search-criterion* だけが使用できます。

<i>descriptor</i>	$\left\{ \begin{array}{l} \text{EQ} \\ = \\ \text{EQUAL} \\ [\text{TO}] \end{array} \right\}$	<i>value</i>
-------------------	---	--------------

検索条件の結合 - DL/I ファイル使用時

[NOT]	$\left\{ \begin{array}{l} \textit{basic-search-criterion} \\ (\textit{search-expression}) \end{array} \right\} \left[ \left\{ \begin{array}{l} \text{OR} \\ \text{AND} \end{array} \right\} \textit{search-expression} \right] \dots$
-------	---

異なるセグメントタイプを参照する *basic-search-criteria* は、OR 論理演算子で結合しないでください。

例：

```
FIND COURSE WITH COURSEN > 1
FIND COURSE WITH COURSEN > 1 AND COURSEN < 100
FIND OFFERING WITH (COURSEN-COURSE > 1 OR TITLE-COURSE = 'Natural')
AND LOCATION = 'DARMSTADT'
```


無効な例：

```
FIND OFFERING WITH COURSEN-COURSE > 1 OR LOCATION = 'DARMSTADT'
```

## COUPLED 節

この節は Adabas データベースにのみ適用されます。

この節は Entire System Server では使用できません。



オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
descriptor1	S A	A N P B	不可	不可
descriptor2	S A	A N P B	不可	不可

**注意:** VIA 節を指定しない場合は、COUPLED 節を 4 回まで指定できますが、VIA 節を使用すると、42 回まで指定できます。

COUPLED 節は、Adabas のカップリング機能を使用した検索を行います。この機能により、1 つの FIND ステートメントの検索条件の中に、異なるファイルのデータベースディスクリプタを指定することができます。

同じ Adabas ファイルを、1 つの FIND ステートメントの異なる FIND COUPLED 節に使用しないでください。

set-name (**RETAIN 節**参照) を basic-search-criteria に指定しないでください。

COUPLED 節内で指定したファイルのデータベースフィールドは、別の FIND や READ ステートメントをカップリングされたファイルに発行しない限り、プログラム内で後から参照することはできません。

**注意:** COUPLED 節を使用する場合は、メインの WITH 節を省略できます。メインの WITH 節を省略するときは、COUPLED 節でキーワード AND/OR を指定しないでください。

### 物理カップリング (VIA 指定なし)

VIA を指定しない COUPLED 節に使用するファイルは、所定の Adabas ユーティリティ (Adabas ドキュメントを参照) を使用して物理的にカップリングされている必要があります。

「例 7 - 物理カップリングファイルの使用」も参照してください。



上の例で、DISPLAY ステートメントの NAME は、EMPLOYEES ファイル内にあるので参照できますが、MAKE は COUPLED 節に指定された VEHICLES ファイル内にあるので参照できません。

EMPLOYEES と VEHICLES が物理的にカップリングされている場合にのみ、レコードの検索が行われます。

### 論理カップリング - VIA 節

オプション `VIA descriptor1 = descriptor2` により、検索参照で複数の Adabas ファイルの論理カップリングが可能となります。

- `descriptor1` は最初のビューのフィールドであり、
- `descriptor2` は 2 番目のビューのフィールドです。

2つのファイルは Adabas で物理的にカップリングする必要はありません。この COUPLED オプションは、Adabas のソフトカップリング機能 (Adabas ドキュメントを参照) を使用します。


「[例 8 - VIA 節](#)」も参照してください。

### STARTING WITH 節

この節は、Adabas および VSAM データベースにのみ適用されます。VSAM では、ESDS にのみ有効です。

`operand5` として、Adabas ISN (内部シーケンス番号) または VSAM RBA (相対バイトアドレス) を指定するために、この節を使用すると、レコードの選択のための開始値として使用されます。

この節は、処理を継続する次のレコードを簡単に確定するために、処理が中断された FIND ループ内で再位置決めで使用できます。これは特に、次のレコードをディスクリプタ値で一意に識別できない場合に役立ちます。また、分散クライアント/サーバーアプリケーションで、レコードをサーバープログラムで読み取ってクライアントプログラムで処理し、レコードを順次処理ではなくバッチ処理する場合にも役立ちます。

 **注意:** 実際に使用される開始値は `operand5` の値ではなく、次に高い値になります。

例:

ライブラリ SYSEXSYN のプログラム FNDSISN を参照してください。

## SORTED BY 節


この節は、Adabas、または SQL データベースにのみ適用されます。

この節は Entire System Server では使用できません。

```
SORTED [BY] descriptor ... 3 [DESCENDING]
```

SORTED BY 節は、選択したレコードを 1 個～3 個のディスクリプタを基準に Adabas ソートします。ソートの順番を制御するディスクリプタと、レコードの選択に使用するディスクリプタは、異なるものを指定できます。

デフォルトで、レコードは昇順にソートされます。降順にソートする場合は、キーワード DESCENDING を指定します。ソートは、レコードを読み込まず、Adabas インバーテッドリストを使用して行われます。

 **注意:** ソート順を制御するディスクリプタの値の数が多くなると負荷も大きくなります。これは、選択された全レコードがリスト内に位置づけられるまで値のリスト全体をスキャンするからです。大量のレコードをソートするときには、SORT ステートメントを使用してください。

Adabas のソート制限 (Adabas ドキュメントの ADARUN の LS パラメータを参照) は、SORTED BY 節を使用するときに影響します。

PE 内のディスクリプタを SORTED BY 節に使用しないでください。MU (添字指定なし) は指定できません。

SORTED BY 節にディスクリプタ以外も指定できます。ただし、メインフレーム環境では、この機能は利用できません。

SORTED BY 節を使用する場合、RETAIN 節と一緒に使用しないでください。

「[例 9 - SORTED BY 節](#)」も参照してください。

## STARTING WITH 節と SORTED BY 節を組み合わせて使用する場合の考慮事項

STARTING WITH 節と SORTED BY 節の両方が同じ FIND ステートメントで使用され、基準データベースが Adabas の場合は、次の点を考慮する必要があります。

### Adabas for Mainframes の場合

Adabas for Mainframes では、FIND ステートメントは次の手順で実行されます。

1. 検索条件に一致するすべてのレコードが収集され、ISN 順序で配置されます。
2. レコードは、SORTED BY 節で指定されたディスクリプタの順にソートされます。

3. ISN 値が STARTING WITH 節で指定されているレコードは、「sorted-by-descriptor」レコードリストに配置されます。
4. 手順 3 で見つかったレコードに続くレコードが、FIND ループで返されます。

### Adabas for OpenSystems の場合

Adabas for OpenSystems (UNIX、OpenVMS、Windows) では、同じステートメントは次のように実行されます。

1. 検索条件に一致するすべてのレコードが収集され、ISN 順序で配置されます。
2. ISN 値が STARTING WITH 節で指定されているレコードは、「sorted-by-ISN」レコードリストに配置されます。
3. 手順 2 で見つかったレコードに続くすべてのレコードは、SORTED BY 節で指定されたディスクリプタ順にソートされ、FIND ループで返されます。

例：

次のプログラムがメインフレーム用の Adabas バージョン 8 および UNIX/OpenVMS/Windows 用の Adabas バージョン 6.1 で実行される場合：

```

DEFINE DATA LOCAL
1 V1 VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 CITY
1 #ISN (I4)
END-DEFINE
FORMAT NL=5 SG=OFF PS=43 AL=15
*
PRINT 'FIND' (I)
FIND V1 WITH NAME = 'B' THRU 'BALBIN'
  RETAIN AS 'SET1'
  IF *COUNTER = 4 THEN
    #ISN := *ISN
  END-IF
  DISPLAY *ISN V1
END-FIND
*
PRINT / 'FIND .. SORTED BY NAME' (I)
FIND V1 WITH 'SET1'
  SORTED BY NAME
  DISPLAY *ISN V1
END-FIND
*
PRINT / 'FIND .. STARTING WITH ISN = ' (I) #ISN (AD=I)
FIND V1 WITH 'SET1'
  STARTING WITH ISN = #ISN
  DISPLAY *ISN V1

```

## FIND

```

END-FIND
*
PRINT / 'FIND .. STARTING WITH ISN = ' (I) #ISN (AD=I)
      ' .. SORTED BY NAME' (I)
FIND V1 WITH 'SET1'
  STARTING WITH ISN = #ISN
  SORTED BY NAME
  DISPLAY *ISN V1
END-FIND
END

```

結果は次のとおりです。

Natural for Mainframes (Adabas バージョン 8) での結果			Natural for OpenSystems (Adabas バージョン 6.1) での結果		
ISN	NAME	FIRST-NAME	ISN	NAME	FIRST-NAME
CITY			CITY		
-----			-----		
-----			-----		
FIND V1 WITH NAME = 'B' THRU 'BALBIN'			FIND V1 WITH NAME = 'B' THRU 'BALBIN'		
12	BAILLET	PATRICK	12	BAILLET	PATRICK
	LEZ LANNOY			LYS LEZ LANNOY	
58	BAGAZJA	MARJAN	58	BAGAZJA	MARJAN
	MONTHERME			MONTHERME	
351	BAECKER	JOHANNES	351	BAECKER	JOHANNES
	FRANKFURT			FRANKFURT	
355	BAECKER	KARL	355	BAECKER	KARL
	SINDELFINGEN			SINDELFINGEN	
370	BACHMANN	HANS	370	BACHMANN	HANS
	MUENCHEN			MUENCHEN	
490	BALBIN	ENRIQUE	490	BALBIN	ENRIQUE
	BARCELONA			BARCELONA	
650	BAKER	SYLVIA	650	BAKER	SYLVIA
	BROOK			OAK BROOK	
913	BAKER	PAULINE	913	BAKER	PAULINE
	DERBY			DERBY	
FIND .. SORTED BY NAME			FIND .. SORTED BY NAME		
370	BACHMANN	HANS	370	BACHMANN	HANS
	MUENCHEN			MUENCHEN	
351	BAECKER	JOHANNES	351	BAECKER	JOHANNES
	FRANKFURT			FRANKFURT	
355	BAECKER	KARL	355	BAECKER	KARL
	SINDELFINGEN			SINDELFINGEN	
58	BAGAZJA	MARJAN			

Natural for Mainframes (Adabas バージョン 8) での結果	Natural for OpenSystems (Adabas バージョン 6.1) での結果
MONTHERME 12 BAILLET           PATRICK           LYS LEZ LANNOY 650 BAKER           SYLVIA            OAK BROOK 913 BAKER           PAULINE DERBY 490 BALBIN           ENRIQUE BARCELONA	58 BAGAZJA           MARJAN MONTHERME 12 BAILLET           PATRICK LYS LEZ LANNOY 650 BAKER           SYLVIA OAK BROOK 913 BAKER           PAULINE DERBY 490 BALBIN           ENRIQUE BARCELONA
FIND .. STARTING WITH ISN = 355  370 BACHMANN        HANS MUENCHEN 490 BALBIN           ENRIQUE BARCELONA 650 BAKER           SYLVIA            OAK BROOK 913 BAKER           PAULINE DERBY	FIND .. STARTING WITH ISN = 355  370 BACHMANN        HANS MUENCHEN 490 BALBIN           ENRIQUE BARCELONA 650 BAKER           SYLVIA OAK BROOK 913 BAKER           PAULINE DERBY
FIND .. STARTING WITH ISN = 355 .. SORTED BY NAME 58 BAGAZJA           MARJAN MONTHERME 12 BAILLET           PATRICK           LYS LEZ LANNOY 650 BAKER           SYLVIA            OAK BROOK 913 BAKER           PAULINE DERBY 490 BALBIN           ENRIQUE BARCELONA	FIND .. STARTING WITH ISN = 355 .. SORTED BY NAME 370 BACHMANN        HANS MUENCHEN 650 BAKER           SYLVIA OAK BROOK 913 BAKER           PAULINE DERBY 490 BALBIN           ENRIQUE BARCELONA

これらのオプション (SORTED BY または STARTING WITH ISN) の1つを指定した FIND ステートメントは、ステートメントが実行されたシステムにかかわらず、同じ順序で同じレコードを返します。一方、両方の節が一緒に使用されている場合、返される結果は、データベースステートメントの処理に使用されている Adabas プラットフォームによって異なります。

したがって、Natural プログラムが複数のプラットフォームで使用されることを意図している場合は、同じ FIND ステートメントで SORTED BY 節と STARTING WITH ISN 節を組み合わせて使用することは避ける必要があります。

## RETAIN 節

この節は Adabas データベースにのみ適用されます。

この節は Entire System Server では使用できません。

RETAIN AS *operand6*

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand6</i>	C S	A	可	不可

構文要素の説明：

<b>RETAIN AS</b>	<p>RETAIN 節を使用すると、大規模ファイルの検索結果を以降の処理のために保持しておくことができます。</p> <p>選択されたレコードは、Adabas ワークファイルに ISN-set として保持されます。この集合を後続の FIND ステートメントの基本検索条件で指定して、さらにしぼり込んだり、レコードを処理したりできます。</p> <p>生成された集合は、特定のファイルに属します。他の FIND ステートメントで集合を使用する場合は、同じファイルを処理する必要があります。集合は、どの Natural プログラムでも参照できます。</p>
<i>operand6</i>	<p>集合名：</p> <p>集合名は、レコードの集合を識別するために使用します。英数字定数あるいは英数字のユーザー定義変数の内容として指定できます。集合名の重複はチェックされません。そのため、同じ集合名を指定すると、新しい集合が古い集合を置き換えます。</p>

「例 10 - RETAIN 節」も参照してください。

### 集合の解放

保持できる集合数や集合内の ISN の件数については、特別な制限はありません。ただし、一度に定義する ISN の集合数は、最小限にとどめておくことをお勧めします。不要になった集合は、RELEASE SETS ステートメントで解放してください。

RELEASE ステートメントで集合を解放しないと、保持された集合は自動的に解放されず、Natural セッション全体を通して存在します。また、別のライブラリにログオンするまで存在します。あるプログラムで作成した集合を、他のプログラムで処理したり、検索条件を追加してさらにしぼり込んだりするために参照できます。

## 他のユーザーによる更新処理

保持された集合内の ISN で識別されたレコードは、他のユーザーのアクセスおよび（または）更新をロックすることはありません。集合からレコードを処理する前に、集合作成時に使用した検索条件が、その時点でも有効であることを確認するには、FIND ステートメントの基本検索条件として WITH 節に集合名を指定し、WHERE 節に元の検索条件（集合作成時に FIND ステートメントで使用した WITH 節の基本検索条件）を指定します。

## 制限事項

RETAIN 節を使用する場合、SORTED BY 節と一緒に使用しないでください。

## WHERE 節

```
WHERE logical-condition
```

WHERE 節は、追加の選択条件 (*logical-condition*) を指定するために使用できます。この条件は、値が読み取られた後、値に対する処理が実行される前に評価されます (AT BREAK 評価を含む)。

*logical-condition* の構文については、『プログラミングガイド』の「論理条件基準」を参照してください。

WHERE 節を持つ FIND ステートメントに、処理数制限の指定がある場合、WHERE 節により除かれたレコードは制限数の合計数から除かれます。しかし、これらのレコードは、Natural セッションパラメータ LT、GLOBALS コマンド、LIMIT ステートメントに指定されたグローバルリミットに対しては、合計されます。

「例 11 - WHERE 節」も参照してください。

## IF NO RECORDS FOUND 節

### ストラクチャードモード構文

```
IF NO [RECORDS][FOUND]
    { ENTER
      statement ... }
END-NOREC
```

## レポートモード構文

```
IF NO [RECORDS] [FOUND]
{
  ENTER
  statement
  DO statement ... DOEND
}
```

IF NO RECORDS FOUND 節を使用すると、WITH 節および WHERE 節で指定された選択条件を満たすレコードがない場合でも、FIND ステートメントで始まる処理ループに入ります。

IF NO RECORDS FOUND 節は、WITH および WHERE 条件の指定を満たすレコードがない場合、1回だけ「空」のレコードの FIND 処理ループを実行します。これが望ましくない場合は、IF NO RECORDS FOUND 節に ESCAPE BOTTOM ステートメントを指定します。

1つ以上のステートメントが IF NO RECORDS FOUND 節で指定されると、処理ループに入る直前にこれらのステートメントが実行されます。ループに入る前に実行するステートメントがない場合は、キーワード ENTER を使用する必要があります。

「例 12 - IF NO RECORDS FOUND 節」も参照してください。

## データベース値

IF NO RECORDS FOUND 節内のステートメントで値が割り当てられない限り、Natural はループで指定されたファイルを参照するすべてのデータベースフィールドをリセットして空にします。

## システム機能の評価

Natural システム機能は、IF NO RECORDS FOUND 節の結果として処理用に作成される空のレコードに対して 1 回評価されます。

## 制限事項

この節は、FIND FIRST、FIND NUMBER、および FIND UNIQUE オプションでは使用できません。

## 例

- 例 1 - PASSWORD 節
- 例 2 - CIPHER 節
- 例 3 - WITH 節の基本検索条件
- 例 4 - マルチプルバリューフィールドの基本検索条件
- 例 5 - WITH 節のさまざまな複合検索式の例
- 例 6 - データベース配列のさまざまな使用例
- 例 7 - 物理カップリングファイルの使用
- 例 8 - VIA 節の使用
- 例 9 - SORTED BY 節



- 例 10 - RETAIN 節
- 例 11 - WHERE 節
- 例 12 - IF NO RECORDS FOUND 節
- 例 13 - FIND ステートメントで使用可能な Natural システム変数
- 例 14 - 複数の FIND ステートメント

FIND NUMBER の例 (**FNDNUM** プログラム) も参照してください。

### 例 1 - PASSWORD 節

```
** Example 'FNDPWD': FIND (with PASSWORD clause)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 PERSONNEL-ID
*
1 #PASSWORD (A8)
END-DEFINE
*
INPUT 'ENTER PASSWORD FOR EMPLOYEE FILE:' #PASSWORD (AD=N)
LIMIT 2
*
FIND EMPLOY-VIEW PASSWORD = #PASSWORD
      WITH NAME = 'SMITH'
  DISPLAY NOTITLE NAME PERSONNEL-ID
END-FIND
*
END
```

プログラム **FNDPWD** の出力：

```
ENTER PASSWORD FOR EMPLOYEE FILE:
```

### 例 2 - CIPHER 節

```
** Example 'FND CIP': FIND (with PASSWORD/CIPHER clause)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 PERSONNEL-ID
*
1 #PASSWORD (A8)
1 #CIPHER (N8)
END-DEFINE
*
LIMIT 2
INPUT 'ENTER PASSWORD FOR EMPLOYEE FILE: ' #PASSWORD (AD=N)
```

```
 / 'ENTER CIPHER KEY FOR EMPLOYEE FILE: ' #CIPHER (AD=N)
*
FIND EMPLOY-VIEW PASSWORD = #PASSWORD
      CIPHER = #CIPHER
      WITH NAME = 'SMITH'
      DISPLAY NOTITLE NAME PERSONNEL-ID
END-FIND
*
END Output of Program FNDCIP:
```

```
ENTER PASSWORD FOR EMPLOYEE FILE:
ENTER CIPHER KEY FOR EMPLOYEE FILE:
```

### 例 3 - WITH 節の基本検索条件

```
FIND STAFF WITH NAME = 'SMITH'
FIND STAFF WITH CITY NE 'BOSTON'
FIND STAFF WITH BIRTH = 610803
FIND STAFF WITH BIRTH = 610803 THRU 610811
FIND STAFF WITH NAME = 'O HARA' OR = 'JONES' OR = 'JACKSON'
FIND STAFF WITH PERSONNEL-ID = 100082 THRU 100100
      BUT NOT 100087 THRU 100095
```

### 例 4 - マルチプルバリュースフィールドの基本検索条件

基本検索条件に使用するディスクリプタが MU のとき、基本的に 4 種類の異なる結果を取得できます。次の例では、MU-FIELD は MU とします。

```
FIND XYZ-VIEW WITH MU-FIELD = 'A'
```

MU-FIELD のいずれかのオカレンスで A 値を持っているレコードが選択されます。

```
FIND XYZ-VIEW WITH MU-FIELD NOT EQUAL 'A'
```

MU-FIELD のいずれかのオカレンスで A 以外の値を持っているレコードが選択されます。

```
FIND XYZ-VIEW WITH NOT MU-FIELD NOT EQUAL 'A'
```

MU-FIELD のすべてのオカレンスで A 値を持っているレコードが選択されます。

```
FIND XYZ-VIEW WITH NOT MU-FIELD = 'A'
```

MU-FIELD のすべてのオカレンスで A 以外の値を持っているレコードが選択されます。

### 例 5 - WITH 節のさまざまな複合検索式の例

```
FIND STAFF WITH BIRTH LT 19770101 AND DEPT = 'DEPT06'
```

```
FIND STAFF WITH JOB-TITLE = 'CLERK TYPIST'
                AND (BIRTH GT 19560101 OR LANG = 'SPANISH')
```

```
FIND STAFF WITH JOB-TITLE = 'CLERK TYPIST'
                AND NOT (BIRTH GT 19560101 OR LANG = 'SPANISH')
```

```
FIND STAFF WITH DEPT = 'ABC' THRU 'DEF'
                AND CITY = 'WASHINGTON' OR = 'LOS ANGELES'
                AND BIRTH GT 19360101
```

```
FIND CARS WITH MAKE = 'VOLKSWAGEN'
                AND COLOR = 'RED' OR = 'BLUE' OR = 'BLACK'
```

### 例 6 - データベース配列のさまざまな使用例

次の例では、SALARY を PE に属するディスクリプタ、LANG を MU であるとします。

```
FIND EMPLOYEES WITH SALARY LT 20000
```

SALARY の全オカレンスを検索します。

```
FIND EMPLOYEES WITH SALARY (1) LT 20000
```

SALARY の第 1 オカレンスだけを検索します。

```
FIND EMPLOYEES WITH SALARY (1:4) LT 20000 /* invalid
```

検索条件に使用されている PE 内のフィールドに範囲を指定しないでください。

```
FIND EMPLOYEES WITH LANG = 'FRENCH'
```

LANG のすべての値を検索します。

```
FIND EMPLOYEES WITH LANG (1) = 'FRENCH' /* invalid
```

検索条件で使用されている MU に添字を指定しないでください。

## 例 7 - 物理カップリングファイルの使用

```
** Example 'FNDCPL': FIND (using coupled files)
** NOTE: Adabas files must be physically coupled when using the
**       COUPLED clause without the VIA clause.
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
1 VEHIC-VIEW VIEW OF VEHICLES
  2 MAKE
END-DEFINE
*
FIND EMPLOY-VIEW WITH CITY = 'FRANKFURT'
  AND COUPLED TO
  VEHIC-VIEW WITH MAKE = 'VW'
  DISPLAY NOTITLE NAME
END-FIND
*
END
```

## 例 8 - VIA 節の使用

```
** Example 'FNDVIA': FIND (with VIA clause)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
1 VEHIC-VIEW VIEW OF VEHICLES
  2 PERSONNEL-ID
END-DEFINE
*
FIND EMPLOY-VIEW WITH NAME = 'ADKINSON'
  AND COUPLED TO VEHIC-VIEW
  VIA PERSONNEL-ID = PERSONNEL-ID WITH MAKE = 'VOLVO'
  DISPLAY PERSONNEL-ID NAME FIRST-NAME
END-FIND
```

```
*
END
```

プログラム **FNDVIA** の出力：

```
Page          1                                05-01-17  13:18:22
PERSONNEL      NAME                FIRST-NAME
  ID
-----
20011000  ADKINSON                BOB
```

### 例9 - SORTED BY 節

```
** Example 'FNDSOR': FIND (with SORTED BY clause)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 NAME
  2 FIRST-NAME
  2 PERSONNEL-ID
END-DEFINE
*
LIMIT 10
FIND EMPLOY-VIEW WITH CITY = 'FRANKFURT'
      SORTED BY NAME PERSONNEL-ID

  DISPLAY NOTITLE NAME (IS=ON) FIRST-NAME PERSONNEL-ID
END-FIND
*
END
```

プログラム **FNDSOR** の出力：

```
      NAME                FIRST-NAME    PERSONNEL
      ID
-----
BAECKER                JOHANNES     11500345
BECKER                 HERMANN      11100311
BERGMANN              HANS         11100301
BLAU                  SARAH        11100305
BLOEMER              JOHANNES     11200312
DIEDRICHS            HUBERT       11600301
DOLLINGER            MARGA        11500322
FALTER               CLAUDIA      11300311
FREI                  HEIDE        11400311
FREI                  REINHILD     11500301
```

**例 10 - RETAIN 節**

```

** Example 'RELEX1': FIND (with RETAIN clause and RELEASE statement)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 BIRTH
  2 NAME
*
1 #BIRTH (D)
END-DEFINE
*
MOVE EDITED '19400101' TO #BIRTH (EM=YYYYMMDD)
*
FIND NUMBER EMPLOY-VIEW WITH BIRTH GT #BIRTH
  RETAIN AS 'AGESET1'
IF *NUMBER = 0
  STOP
END-IF
*
FIND EMPLOY-VIEW WITH 'AGESET1' AND CITY = 'NEW YORK'
  DISPLAY NOTITLE NAME CITY BIRTH (EM=YYYY-MM-DD)
END-FIND
*
RELEASE SET 'AGESET1'
*
END

```

例 10 の出力：

NAME	CITY	DATE OF BIRTH
RUBIN	NEW YORK	1945-10-27
WALLACE	NEW YORK	1945-08-04

**例 11 - WHERE 節**

```

** Example 'FNDWHE': FIND (with WHERE clause)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 JOB-TITLE
  2 CITY
END-DEFINE

```

```

*
FIND EMPLOY-VIEW WITH CITY = 'PARIS'
      WHERE JOB-TITLE = 'INGENIEUR COMMERCIAL'
      DISPLAY NOTITLE
      CITY JOB-TITLE PERSONNEL-ID NAME
END-FIND
*
END

```

プログラム **FNDWHE** の出力：

CITY	CURRENT POSITION	PERSONNEL ID	NAME
PARIS	INGENIEUR COMMERCIAL	50007300	CAHN
PARIS	INGENIEUR COMMERCIAL	50006500	MAZUY
PARIS	INGENIEUR COMMERCIAL	50004700	FAURIE
PARIS	INGENIEUR COMMERCIAL	50004400	VALLY
PARIS	INGENIEUR COMMERCIAL	50002800	BRETON
PARIS	INGENIEUR COMMERCIAL	50001000	GIGLEUX
PARIS	INGENIEUR COMMERCIAL	50000400	KORAB-BRZOZOWSKI

## 例 12 - IF NO RECORDS FOUND 節

```

** Example 'FNDIFN': FIND (using IF NO RECORDS FOUND)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
1 VEHIC-VIEW VIEW OF VEHICLES
  2 PERSONNEL-ID
  2 MAKE
END-DEFINE
*
LIMIT 15
EMP. READ EMPLOY-VIEW BY NAME STARTING FROM 'JONES'
/*
  VEH. FIND VEHIC-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (EMP.)

  IF NO RECORDS FOUND
    MOVE '*** NO CAR ***' TO MAKE
  END-NOREC
/*
  DISPLAY NOTITLE
    NAME (EMP.) (IS=ON)
    FIRST-NAME (EMP.) (IS=ON)
    MAKE (VEH.)
END-FIND

```

## FIND

```
/*  
END-READ  
END
```

プログラム **FNDIFN** の出力：

NAME	FIRST-NAME	MAKE
JONES	VIRGINIA	CHRYSLER
	MARSHA	CHRYSLER
		CHRYSLER
	ROBERT	GENERAL MOTORS
	LILLY	FORD
		MG
	EDWARD	GENERAL MOTORS
	MARTHA	GENERAL MOTORS
	LAUREL	GENERAL MOTORS
	KEVIN	DATSUN
GREGORY	FORD	
JOPER	MANFRED	*** NO CAR ***
JOUSSELIN	DANIEL	RENAULT
JUBE	GABRIEL	*** NO CAR ***
JUNG	ERNST	*** NO CAR ***
JUNKIN	JEREMY	*** NO CAR ***
KAISER	REINER	*** NO CAR ***

### 例 13 - FIND ステートメントで使用可能な Natural システム変数

```
** Example 'FNDVAR': FIND (using *ISN, *NUMBER, *COUNTER)  
*****  
DEFINE DATA LOCAL  
1 EMPLOY-VIEW VIEW OF EMPLOYEES  
  2 PERSONNEL-ID  
  2 NAME  
  2 CITY  
END-DEFINE  
*  
LIMIT 3  
FIND EMPLOY-VIEW WITH CITY = 'MADRID'  
  DISPLAY NOTITLE PERSONNEL-ID NAME  
          *ISN *NUMBER *COUNTER  
END-FIND  
*  
END
```



## プログラム **FNDVAR** の出力

PERSONNEL ID	NAME	ISN	NMBR	CNT
60000114	DE JUAN	400	41	1
60000136	DE LA MADRID	401	41	2
60000209	PINERO	405	41	3

## 例 14 - 複数の FIND ステートメント

EMPLOYEES ファイルを FIND し、姓が SMITH である人をすべて選択します。この EMPLOYEES ファイルの PERSONNEL-ID を検索キーにして、VEHICLES ファイルをアクセスします。

```
** Example 'FNDMUL': FIND (with multiple files)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
1 VEHIC-VIEW VIEW OF VEHICLES
  2 PERSONNEL-ID
  2 MAKE
END-DEFINE
*
LIMIT 15
EMP. FIND EMPLOY-VIEW WITH NAME = 'SMITH'
/*
  VEH. FIND VEHIC-VIEW WITH PERSONNEL-ID = EMP.PERSONNEL-ID
  IF NO RECORDS FOUND
    MOVE '*** NO CAR ***' TO MAKE
  END-NOREC
  DISPLAY NOTITLE
    EMP.NAME (IS=ON)
    EMP.FIRST-NAME (IS=ON)
    VEH.MAKE
  END-FIND
END-FIND
END
```

## プログラム **FNDMUL** の出力：

レポートに表示されるのは、姓が SMITH であるすべての人の NAME と FIRST-NAME (姓名) (EMPLOYEES ファイルより入手) および彼らの所有する車の MAKE (型) (VEHICLES ファイルより入手) です。

# FIND

---

NAME	FIRST-NAME	MAKE
SMITH	GERHARD	ROVER
	SEYMOUR	*** NO CAR ***
	MATILDA	FORD
	ANN	*** NO CAR ***
	TONI	TOYOTA
	MARTIN	*** NO CAR ***
	THOMAS	FORD
	SUNNY	*** NO CAR ***
	MARK	FORD
	LOUISE	CHRYSLER
	MAXWELL	MERCEDES-BENZ
		MERCEDES-BENZ
	ELSA	CHRYSLER
	CHARLY	CHRYSLER
	LEE	*** NO CAR ***
FRANK	FORD	

# 65 FOR

---

■ 機能 .....	442
■ 構文説明 .....	442
■ 例 .....	443

FOR operand1	[	{	[:] =	}	operand2
			EQ		
			FROM		
			TO		operand3
			THRU		[[STEP] operand4]
					statement ...
END-FOR	(structured mode only)				
[LOOP]	(reporting mode only)				

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[REPEAT](#) | [ESCAPE](#)

関連機能グループ：「[ループ実行](#)」

## 機能

FORステートメントは、処理ループを開始し、ループの処理回数を制御するために使用します。

### 整合性チェック

FORループに入る前に、オペランドの値に矛盾がないか（operand2にoperand4を繰り返し加算し、operand3の値以上になるか）を確認めます。値に矛盾があれば、FORループには入りません（エラーメッセージは出力されません）。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素				フォーマット											ステートメント参照	ダイナミック定義		
operand1	S				N	P	I	F										可	可
operand2	C	S		N	N	P	I	F										可	不可
operand3	C	S		N	N	P	I	F										可	不可
operand4	C	S		N	N	P	I	F										可	不可

構文要素の説明：

<i>operand1</i>	ループ制御変数 ( <i>operand1</i> ) および初期設定 ( <i>operand2</i> )
<i>operand2</i>	<p><i>operand1</i>は、処理ループの実行回数を制御します。データベースフィールドまたはユーザー定義変数が使用できます。キーワード FROM の後の指定値 (<i>operand2</i>) が、最初に処理ループに入る前に制御変数フィールドに割り当てられます。制御変数フィールドの値は、ループに入るたびにキーワード STEP の後の指定値 (<i>operand4</i>) ずつ増えていきます (STEP 値が負数であれば減っていきます)。</p> <p>処理ループ実行中にループ制御変数を参照できます。ループ制御変数には、その現在値が含まれます。</p>
<i>operand3</i>	<p><b>TO 値：</b></p> <p>処理ループは、<i>operand1</i>が <i>operand3</i>の値よりも大きくなると (STEP 値が負であれば小さくなると) 終了します。</p>
<i>operand4</i>	<p><b>STEP 値：</b></p> <p>STEP 値には正でも負でも指定できます。STEP 値を省略すると、+1 の増加とみなされます。</p> <p>ループに入るときに STEP 値の符号によって、「より小さい」または「より大きい」の比較演算が行われます。</p> <p><i>operand4</i> をゼロにすることはできません。</p>
<b>END-FOR</b>	FOR ステートメントを終了するには、Natural の予約語 END-FOR を使用する必要があります。

## 例

```

** Example 'FOREX1S': FOR (structured mode)
*****
DEFINE DATA LOCAL
1 #INDEX (I1)
1 #ROOT (N2.7)
END-DEFINE
*
FOR #INDEX 1 TO 5
  COMPUTE #ROOT = SQRT (#INDEX)
  WRITE NOTITLE '=' #INDEX 3X '=' #ROOT
END-FOR
*
SKIP 1
FOR #INDEX 1 TO 5 STEP 2
  COMPUTE #ROOT = SQRT (#INDEX)
  WRITE '=' #INDEX 3X '=' #ROOT
END-FOR
*
END

```

プログラム **FOREX1S** の出力：

```
#INDEX: 1 #ROOT: 1.0000000
#INDEX: 2 #ROOT: 1.4142135
#INDEX: 3 #ROOT: 1.7320508
#INDEX: 4 #ROOT: 2.0000000
#INDEX: 5 #ROOT: 2.2360679

#INDEX: 1 #ROOT: 1.0000000
#INDEX: 3 #ROOT: 1.7320508
#INDEX: 5 #ROOT: 2.2360679
```

レポートモードの例：**FOREX1R**。

# 66      FORMAT

---

▪ 機能 .....	446
▪ 構文説明 .....	446
▪ 適用可能なパラメータ .....	447
▪ 例 .....	448

FORMAT [(*rep*)] *parameter* ...

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT END OF PAGE](#) | [AT TOP OF PAGE](#) | [CLOSE PRINTER](#) | [DEFINE PRINTER](#) | [DISPLAY](#) | [EJECT](#) | [NEWPAGE](#) | [PRINT](#) | [SKIP](#) | [SUSPEND IDENTICAL SUPPRESS](#) | [WRITE](#) | [WRITE TITLE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

FORMAT ステートメントは、入出力時のパラメータ設定を指定します。

FORMAT ステートメントで指定した設定は、GLOBALS コマンドや SET GLOBALS ステートメント（レポートモードのみ）、または Natural 管理者が設定した、セッションに有効なデフォルトの設定値を上書きします（コンパイル時）。

[DISPLAY](#)、[INPUT](#)、[PRINT](#)、[WRITE](#)、[WRITE TITLE](#)、または [WRITE TRAILER](#) ステートメントでパラメータを指定すると、この設定値も上書きされます。

設定値は、プログラムの終了または次の FORMAT ステートメントが現れるまで有効です。

FORMAT ステートメントは、Natural 実行コードを生成するものではありません。実行はプログラムの論理的な処理の流れに依存しません。このステートメントは、プログラムのコンパイル時に評価されます。つまり、[DISPLAY](#)、[WRITE](#)、[PRINT](#)、および [INPUT](#) ステートメントのコンパイル時のパラメータを設定します。FORMAT ステートメントで定義された設定は、それに続くすべての [DISPLAY](#)、[WRITE](#)、[PRINT](#)、および [INPUT](#) ステートメントに適用されます。

## 構文説明

<i>(rep)</i>	<p>レポート指定：</p> <p>表記 (<i>rep</i>) は、FORMAT ステートメントを適用するレポートの ID を指定するために使用できます。</p> <p>範囲 0~31 の値、または <a href="#">DEFINE PRINTER</a> ステートメントを使用して割り当てた論理名を指定できます。</p> <p>(<i>rep</i>) を指定しない場合、FORMAT ステートメントは最初のレポート（レポート 0）に適用されます。</p> <p>Natural で作成される出力レポートの形式を制御する方法については、『<a href="#">プログラミングガイド</a>』の「<a href="#">データ出力制御</a>」を参照してください。</p>
--------------	---




<i>parameter</i>	<p>パラメータ：</p> <p>パラメータの指定順序は任意で、1つ以上の空白で区切って指定する必要があります。1つのエントリを2つのステートメント行に分けて入力しないでください。</p> <p>ここに適用されるフィールド対応パラメータ設定は、選択されたレポートのINPUT、WRITE、DISPLAY、またはPRINTステートメントで使用される変数フィールドにのみ関連します。これらのステートメントに使用される文字定数には適用しません。</p> <p>例：</p> <pre> DEFINE DATA LOCAL 1 VARI (A4)      INIT &lt;'1234'&gt;      /*      Output END-DEFINE      /*      Produced FORMAT AD=U      /*      ----- WRITE 'Text'      VARI              /*      Text 1234 WRITE 'Text' (AD=U) VARI            /*      Text1234 END </pre> <p>下記の「アプリケーションパラメータ」も参照してください。</p>
------------------	--

## 適用可能なパラメータ

使用できるセッションパラメータの詳細な説明については、『パラメータリファレンス』を参照してください。

パラメータ	説明
AD	属性定義
AL	出力の英数字長
BX	ボックス定義
CD	カラー定義
DF	日付フォーマット
DL	出力の表示長
EM	編集マスク
ES	空行の省略
FC	充填文字
FL	浮動小数点仮数長
GC	グループに対するヘッダー充填（フィラー）文字
HC	ヘッダーの中央揃え
HW	ヘッダーの幅
IC	挿入文字

パラメータ	説明
IP	入力プロンプトのテキスト
IS	重複抑制
KD	キー定義
LC	先頭文字
LS	行サイズ
MC	マルチプルバリューフィールド数
MP	レポートの最大ページ数（下記の「注」を参照）
MS	手動による省略
NL	出力の数値長
PC	ピリオディックグループ数
PM	出力モード
PS	ページサイズ（下記の「注」を参照）
SF	フィールド間の空白
SG	符号の位置
TC	末尾文字
UC	下線付き文字
ZP	ゼロ出力

 **注意:** パラメータ MP および PS は、特定の I/O ステートメントに対しては無効で、レポートに作成された出力全体に適用されます。MP および PS の複数の設定が実行された場合は、最後の定義が使用されます。

『プログラミングガイド』の「タイトルおよびヘッダーの下線付き文字-UC パラメータ」も参照してください。

## 例

```

** Example 'FMTEX1': FORMAT
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CITY
  2 POST-CODE
  2 COUNTRY
END-DEFINE
*
FORMAT AL=7      /* Alpha-numeric field output length
          FC=+    /* Filler character for field header
          GC=*    /* Filler character for group header
    
```

```
HC=L /* Header left justified
IC=<< /* Insert characters
IS=ON /* Identical suppress on
TC=>> /* Trailing character
UC== /* Underline character
ZP=OFF /* Zero print off
*
LIMIT 5
READ EMPLOY-VIEW BY NAME
  DISPLAY NOTITLE
    NAME 3X CITY 3X POST-CODE 3X COUNTRY
END-READ
*
END
```

プログラム **FMTEX1** の出力：

NAME++++++	CITY++++++	POSTAL+++++	COUNTRY++++
=====	=====	=====	=====
<<ABELLAN>>	<<MADRID >>	<<28014 >>	<<E >>
<<ACHIESO>>	<<DERBY >>	<<DE3 4TR>>	<<UK >>
<<ADAM >>	<<JOIGNY >>	<<89300 >>	<<F >>
<<ADKINSO>>	<<BROOKLY>>	<<11201 >>	<<USA>>
	<<BEVERLE>>	<<90211 >>	



# 67 GET

---

▪ 機能 .....	452
▪ 制限 .....	452
▪ 構文説明 .....	453
▪ 例 .....	454

```
GET [IN] [FILE] view-name
  [PASSWORD=operand1]
  [CIPHER=operand2]
  [ { [RECORD] } ] [ { operand3 } operand4
  [ { [RECORDS] } ] [ { *ISN [r] } ] ...
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET SAME](#) | [GET TRANSACTION](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

---

GET ステートメントは、Adabas ISN（内部シーケンス番号）を指定してレコードを読み込みます。

GET ステートメントで処理ループは発生しません。

## 制限

---

- DL/I および SQL データベースに対して GET ステートメントは使用できません。
- Entire System Server では、GET ステートメントを使用できません。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A	可	不可
<i>operand2</i>	C S	N	不可	不可
<i>operand3</i>	C S	N N P I B *	可	不可
<i>operand4</i>	S A	A N P I F B D T L	可	可

\**operand3* のフォーマット B は、4 以下の長さでのみ使用できます。

構文要素の説明：

<i>view-name</i>	ビュー名：  DEFINE DATA ステートメント内またはグローバルデータエリアやローカルデータエリアで定義されたビュー名です。  レポートモードでは、 <i>view-name</i> を DDM の名前にすることもできます。
<b>PASSWORD=</b> <i>operand1</i>	Adabas または VSAM データベースにのみ適用できます。
<b>CIPHER=</b> <i>operand2</i>	PASSWORD 節は、パスワード保護されている Adabas ファイルからデータを読み込むときにパスワードを指定します。  CIPHER 節は、サイファリング (暗号化) されている Adabas ファイルからデータを読み込むとき、サイファキーを指定するために使用します。  詳細については、ステートメント <b>FIND</b> および <b>PASSW</b> を参照してください。
* <b>ISN</b> / <i>operand3</i>	内部シーケンス番号：  ISN は、数値定数、ユーザー定義変数 ( <i>operand3</i> ) の形式、または Natural システム変数 *ISN によって提供されています。  <b>注意:</b> VSAM データベースの場合：VSAM ESDS データセットでは、RBA をユーザー定義変数 (数字フォーマット) または整数の定数として指定する必要があります。VSAM RRDS については、RBA の代わりに RRN を指定することを除いて同じ規則が適用されます。
( <i>r</i> )	ステートメント参照：  表記 ( <i>r</i> ) は、レコードの最初の読み取りに使用される <b>FIND</b> または <b>READ</b> ステートメントを含むステートメントを指定するために使用します。  ( <i>r</i> ) 指定がないと、GETSAME ステートメントは稼働している最も内側の処理ループに関連します。

	(r) は、参照ステートメント番号またはステートメントラベルとして指定できません。
<i>operand4</i>	データベースフィールドの参照：  GET ステートメントで読み込んだフィールドを後で参照する場合、GET ステートメントのラベルまたは行番号を指定する必要があります。  <i>operand4</i> は、ストラクチャードモードでは使用できません。

## 例

```

** Example 'GETEX1': GET
*****
DEFINE DATA LOCAL
1 PERSONS VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
1 SALARY-INFO VIEW OF EMPLOYEES
  2 NAME
  2 CURR-CODE (1:1)
  2 SALARY    (1:1)
*
1 #ISN-ARRAY (B4/1:10)
1 #LINE-NR   (N2)
END-DEFINE
*
FORMAT PS=16
LIMIT 10
READ PERSONS BY NAME
  MOVE *COUNTER TO #LINE-NR
  MOVE *ISN     TO #ISN-ARRAY (#LINE-NR)
  DISPLAY #LINE-NR PERSONNEL-ID NAME FIRST-NAME
  /*
  AT END OF PAGE
  INPUT / 'PLEASE SELECT LINE-NR FOR SALARY INFORMATION:' #LINE-NR
  IF #LINE-NR = 1 THRU 10
    GET SALARY-INFO #ISN-ARRAY (#LINE-NR)
    WRITE / SALARY-INFO.NAME
           SALARY-INFO.SALARY    (1)
           SALARY-INFO.CURR-CODE (1)
  END-IF
END-ENDPAGE
/*
END-READ
END

```



プログラム **GETEX1** の出力：

Page 1 05-01-13 13:17:42

#LINE-NR	PERSONNEL ID	NAME	FIRST-NAME
----------	-----------------	------	------------

1	60008339	ABELLAN	KEPA
2	30000231	ACHIESON	ROBERT
3	50005800	ADAM	SIMONE
4	20008800	ADKINSON	JEFF
5	20009800	ADKINSON	PHYLLIS
6	20012700	ADKINSON	HAZEL
7	20013800	ADKINSON	DAVID
8	20019600	ADKINSON	CHARLIE
9	20008600	ADKINSON	MARTHA
10	20005700	ADKINSON	TIMMIE

PLEASE SELECT LINE-NR FOR SALARY INFORMATION: 1

ABELLAN 1450000 PTA



# 68 GET SAME

---

▪ 機能 .....	458
▪ 制限 .....	458
▪ 構文説明 .....	459
▪ 例 .....	459

ストラクチャードモード構文

```
GET SAME [(r)]
```

レポートモード構文

```
GET SAME [(r)] [operand1 ...]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET TRANSACTION DATA](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

---

GET SAME ステートメントは、現在処理中のレコードを再度読み込むために使用します。このステートメントは、最初にレコードを読む時点で、存在するオカレンスまたは使用するオカレンスの数や範囲がわからない場合に、データベース配列の値（PE または MU）を参照するために最もよく使用します。

## 制限

---

- GET SAME ステートメントは Adabas または VSAM を使用している Natural ユーザーに対してのみ有効です。
- GET SAME ステートメントは Entire System Server では使用できません。
- VSAM データベースの場合、GET SAME は ESDS と RRDS にのみ適用されます。ESDS に対しては、RBA をユーザー定義変数（数字フォーマット）または整数として指定する必要があります。RBA の代わりに RRN を指定することを除けば、RRDS にも同じ規則が適用されます。
- UPDATE または DELETE ステートメントでは GET SAME ステートメントを参照しないでください。これらのステートメントは、最初にレコードを読むときに使用した FIND、READ、または GET ステートメントを参照します。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	動的定義
<i>operand1</i>	S A	A U N P B	不可	可

構文要素の説明：

<i>(r)</i>	<p>ステートメント参照：</p> <p>表記 <i>(r)</i> は、レコードの最初の読み取りに使用される <b>FIND</b> または <b>READ</b> ステートメントを含むステートメントを指定するために使用します。</p> <p><i>(r)</i> 指定がないと、GET SAME ステートメントは稼働している最も内側の処理ループに関連します。</p> <p><i>(r)</i> は、参照ステートメント番号またはステートメントラベルとして指定できます。</p>
<i>operand1</i>	<p>参照可能になるフィールド：</p> <p><i>operand1</i>として、GET SAME ステートメントの結果、参照可能になるフィールドを指定します。</p> <p><b>注意:</b> フィールドを <b>DEFINE DATA</b> ステートメントで定義している場合、<i>operand1</i>は指定できません。</p>

## 例

```
** Example 'GSAEX1': GET SAME
*****
DEFINE DATA LOCAL
1 I          (P3)
1 POST-ADDRESS VIEW OF EMPLOYEES
  2 FIRST-NAME
  2 NAME
  2 ADDRESS-LINE (I:I)
  2 C*ADDRESS-LINE
  2 POST-CODE
  2 CITY
*
1 #NAME      (A30)
END-DEFINE
*
```

## GET SAME

---

```
FORMAT PS=20
MOVE 1 TO I
*
READ (10) POST-ADDRESS BY NAME
  COMPRESS NAME FIRST-NAME INTO #NAME WITH DELIMITER ','
  WRITE // 12T #NAME
  WRITE / 12T ADDRESS-LINE (I.1)
  /*
  IF C*ADDRESS-LINE > 1
    FOR I = 2 TO C*ADDRESS-LINE
      GET SAME /* READ NEXT OCCURRENCE
      WRITE 12T ADDRESS-LINE (I.1)
    END-FOR
  END-IF
  WRITE / POST-CODE CITY
  SKIP 3
END-READ
END
```

プログラム **GSAEX1** の出力：

Page 1 05-01-13 13:23:36

```
ABELLAN,KEPA
CASTELAN 23-C
28014 MADRID
```

```
ACHIESON,ROBERT
144 ALLESTREE LANE
DERBY
DERBYSHIRE
DE3 4TR DERBY
```

# 69

## GET TRANSACTION DATA

---

▪ 機能 .....	462
▪ 制限事項 .....	463
▪ 構文説明 .....	463
▪ 例 .....	463

```
GET TRANSACTION [DATA] operand1 ...
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

---

GET TRANSACTION DATA ステートメントは、以前に [END TRANSACTION](#) ステートメントで保存したデータを読むために使用します。

GET TRANSACTION DATA は、処理ループを開始しません。



**注意:** DL/I の場合：GET TRANSACTION DATA ステートメントは、END TRANSACTION ステートメントで保存したチェックポイントデータを取得します。

### システム変数 \*ETID

Natural システム変数 \*ETID は、データベースから読み込むトランザクションデータを識別します。

### トランザクションデータが存在しない場合

GET TRANSACTION DATA ステートメントが発行されたときに、トランザクションデータが存在しない場合、GET TRANSACTION DATA ステートメントで指定した全フィールドは、定義されたフォーマットにかかわらず全桁が空白で返されます。



**注意:** 空のデータに対し算術演算を行うとプログラムは異常終了します。そのため、ユーザーは、「空」のトランザクションデータに対して算術演算を行わないように確認する必要があります。



## 制限事項

GET TRANSACTION DATA ステートメントは、Adabas データベースまたはバッチ関連 BMP リージョンの DL/I データベース（IMS 環境）にのみ有効です。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S	A U N P I F B D T	可	可

構文要素の説明：

<i>operand1</i>	<p>フィールド指定：</p> <p>GET TRANSACTION DATA ステートメントに使用するフィールドの順序、長さ、およびフォーマットは、対応する END TRANSACTION ステートメントで指定したフィールドの順序、長さ、およびフォーマットと同じにする必要があります。</p> <p><b>注意:</b> DL/I の場合：最初の <i>operand1</i> は 8 バイトのチェックポイント ID である必要があります。</p>
-----------------	--

## 例

```
** Example 'GTREX1': GET TRANSACTION
**
** CAUTION: Executing this example will modify the database records!
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
  2 MIDDLE-I
  2 CITY
*
1 #PERS-NR (A8) INIT <' '>
END-DEFINE
*
```

## GET TRANSACTION DATA

---

```
GET TRANSACTION DATA #PERS-NR
IF #PERS-NR NE ' '
  WRITE 'LAST TRANSACTION PROCESSED FROM PREVIOUS SESSION' #PERS-NR
END-IF
*
REPEAT
  /*
  INPUT 10X 'ENTER PERSONNEL NUMBER TO BE UPDATED:' #PERS-NR
  IF #PERS-NR = ' '
    STOP
  END-IF
  /*
  FIND EMPLOY-VIEW WITH PERSONNEL-ID = #PERS-NR
  IF NO RECORDS FOUND
    REINPUT 'NO RECORD FOUND'
  END-NOREC
  INPUT (AD=M) PERSONNEL-ID (AD=0)
    / NAME
    / FIRST-NAME
    / CITY

  UPDATE
  END TRANSACTION #PERS-NR
END-FIND
  /*
END-REPEAT
END
```

# 70 HISTOGRAM

---

■ 機能 .....	466
■ 制限 .....	467
■ 構文説明 .....	467
■ 例 .....	473

HISTOGRAM	[ ALL (operand1) ]	[MULTI-FETCH-clause][multi-fetch-factor] [IN][FILE] view-name
	[PASSWORD=operand2]	
	[ IN ]	[SEQUENCE]
	{ ASCENDING DESCENDING VARIABLE operand3 DYNAMIC operand3 }	
	[VALUE][FOR][FIELD] operand4	
	[STARTING/ENDING-clause]	
	[WHERE logical-condition]	
	statement ...	
	END-HISTOGRAM (ストラクチャードモードのみ)	
	[LOOP] (レポートモードのみ)	

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION DATA](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

HISTOGRAM ステートメントは、ディスクリプタ、サブディスクリプタ、またはスーパーディスクリプタとして定義されているデータベースフィールドの値を読むために使用します。値は Adabas インバーテッドリストまたは VSAM インデックスから直接読み取られます。

HISTOGRAM ステートメントは、処理ループを開始します。ただし、HISTOGRAM ステートメントで定義したフィールド以外のフィールドを参照することはできません。

『プログラミングガイド』の「[HISTOGRAM ステートメント](#)」も参照してください。



**注意:** SQL データベースの場合：HISTOGRAM は特定の列に同じ値を持つ行の数を返しません。

## 制限

- DL/I データベースまたは Entire System Server では、このステートメントを使用できません。

VSAM データベースに適用される場合、HISTOGRAM ステートメントは KSDS および ESDS に対してのみ有効です。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	N P I B*	可	不可
<i>operand2</i>	C S	A	可	不可
<i>operand3</i>	S	A	可	不可
<i>operand4</i>	S	A N P I F B D T L	不可	不可

\**operand1* のフォーマット B は、4 以下の長さでのみ使用できます。

構文要素の説明：

<i>operand1</i> /ALL	<p>処理ループの制限：</p> <p>HISTOGRAM ステートメントで <i>operand 1</i> を指定することで処理するディスクリプタ値の数は、数値定数 (0~4294967295) またはユーザー定義変数 (整数値を含む) を指定して制限できます。</p> <p>ALL は全ディスクリプタ値が処理されるように強調するためにオプションとして指定できます。</p> <p>このステートメントでは、指定した制限が、LIMIT ステートメントで設定した制限より優先されます。</p> <p>LT パラメータで、より小さな制限が設定された場合、LT 制限が適用されます。</p> <p><b>注意:</b> 4 桁の件数を処理する場合、リーディングゼロを指定します (0nnnn)。これは、Natural が、カッコで囲まれた 4 桁の数をすべてステートメントの参照番号と解釈するためです。<i>operand1</i> は HISTOGRAM ループに入るときに評価されます。<i>operand1</i> の値が HISTOGRAM ループ内で変更されても、その値は処理するレコード数に影響を与えません。</p>
----------------------	---


<i>MULTI-FETCH-clause</i>	以下の「 <b>MULTI-FETCH 節</b> 」を参照してください。
<i>view-name</i>	<p><i>view-name</i> として、<b>DEFINE DATA</b> ステートメント内、またはグローバルデータエリアやローカルデータエリアで定義したビューの名前です。</p> <p>ビュー内には HISTOGRAM ステートメントで使用するフィールドだけを指定します (<i>operand4</i>) 。</p> <p>ビューのフィールドが、添字範囲付きで定義された PE または MU の場合、その範囲の最初のオカレンスだけが HISTOGRAM ステートメントで参照されます。他のすべてのオカレンスは HISTOGRAM ステートメントの実行には関連しません。</p> <p>レポートモードでは、<i>view-name</i> を DDM の名前にすることもできます。</p>
<b>PASSWORD=</b> <i>operand2</i>	<p><b>PASSWORD 節</b>：</p> <p>パスワードで保護された Adabas ファイルを検索する場合は、<b>PASSWORD 節</b> にパスワード (<i>operand2</i>) を指定します。詳細については、ステートメント <b>FIND</b> および <b>PASSW</b> を参照してください。</p>

SEQUENCE	<p><b>SEQUENCE 節：</b></p> <p>この節は、Adabas、VSAM、SQL データベースに対してだけ使用できます。</p> <p>この節を使用して、レコードを昇順または降順のどちらで読み取るかを指定できます。</p> <ul style="list-style-type: none"> <li>■ デフォルトは昇順です。昇順はキーワード ASCENDING を使用して明示的に指定することもできます。</li> <li>■ レコードを降順で読み取る場合は、キーワード DESCENDING を指定します。</li> <li>■ レコードを昇順または降順のどちらで読み取るかを事前に決めずに実行時に決めるには、変数 (<i>operand3</i>) の前に VARIABLE キーワードまたは DYNAMIC キーワードを指定します。<i>operand3</i> は、フォーマット/長さ A1 で、値 A (「昇順」) または D (「降順」) が含まれている必要があります。</li> <li>■ キーワード VARIABLE が指定された場合、HISTOGRAM 処理ループ開始時に読み込み順 (<i>operand3</i> の値) が評価されます。<i>operand3</i> フィールドが HISTOGRAM ループ内で変更されるかどうかにかかわらず、ループが終了するまで読み込み順は変更されません。</li> <li>■ キーワード DYNAMIC が指定された場合、HISTOGRAM 処理ループ内でレコードを取得する前に読み込み順 (<i>operand3</i> の値) が評価されます。あるレコードからレコードまで読み込み順を変更することが可能です。これは、HISTOGRAM ループの任意の位置で読み込み順を昇順から降順 (または、逆) に変更することが可能ということです。</li> </ul> <p>SEQUENCE 節の例：</p> <ul style="list-style-type: none"> <li>■ <a href="#">例2 - 降順レコード読み取りによる HISTOGRAM ステートメント</a></li> <li>■ <a href="#">例3 - 変数順序を使用した HISTOGRAM ステートメント</a></li> </ul>
<i>operand4</i>	<p><b>ディスクリプタ：</b></p> <p><i>operand4</i> として、ディスクリプタ、サブディスクリプタ、スーパーディスクリプタ、またはハイパーディスクリプタが使用できます。</p> <p>ピリオディックグループに含まれるディスクリプタをインデックス付きまたはなしで指定できます。インデックスなしで指定した場合、ディスクリプタは指定した値がいずれかのオカレンスに見つかった場合に選択されます。インデックス付きで指定した場合、レコードは値がインデックスで指定したオカレンス内に見つかった場合にのみ選択されます。インデックスは定数で指定する必要があります。インデックス範囲は使用できません。</p> <p>MU のディスクリプタでは、インデックスを指定しないでください。値の位置に関係なく、値がレコードに存在していればディスクリプタは選択されます。</p>

<p><b>STARTING-ENDING-clause</b></p>	<p><b>STARTING/ENDING 節：</b></p> <p>開始値と終了値は、キーワード STARTING および ENDING（または、THRU）の後に、処理を開始／終了する値を示す定数またはユーザー定義変数を続けて指定できます。</p> <p>詳細については、以下の「<i>Starting/Ending 値の指定</i>」を参照してください。</p>	
<p><b>WHERE <i>logical-condition</i></b></p>	<p><b>WHERE 節：</b></p> <p>WHERE 節は、追加の選択条件 (<i>logical-condition</i>) を指定するために使用できます。この条件は、値が読み取られた後、値に対する処理が実行される前に評価されます (AT BREAK 評価を含む)。</p> <p>WHERE 節で指定するディスクリプタは、HISTOGRAM ステートメントで参照したディスクリプタと同じにする必要があります。HISTOGRAM ステートメントでは、他のフィールドに対する処理を行うことができません。</p> <p><i>logical-condition</i> の構文については、『プログラミングガイド』の「論理条件基準」で説明しています。</p> <p>システム変数</p> <p>Natural システム変数 *ISN、*NUMBER、および *COUNTER が HISTOGRAM ステートメントで使用可能です。</p> <p>*NUMBER および *ISN は、WHERE 節の評価後に設定されます。これらを WHERE 節の論理条件に使用しないでください。</p>	
<p><b>END-HISTOGRAM</b></p>	<p>*NUMBER</p>	<p>システム変数 *NUMBER には、最後に読まれた値のデータベースレコード件数が入ります。</p> <p>SQL データベースについては、『システム変数』ドキュメントの「SQL データベースに対する *NUMBER」を参照してください。</p>
	<p>*ISN</p>	<p>システム変数 *ISN には、最後に読まれたディスクリプタ値を持つ、オカレンス番号が入ります。*ISN は、PE に属さないディスクリプタに対しては常に 0 を返します。</p> <p>*ISN は、SQL および VSAM データベースでは使用できません。</p>
	<p>*COUNTER</p>	<p>システム変数 *COUNTER には、読まれた値の個数の合計が入ります (WHERE 節の評価後)。</p>



## MULTI-FETCH 節

 **注意:** この節は、Adabas または DB2 データベースでのみ使用できます。

```
[ MULTI-FETCH { ON
                OFF
                OF multi-fetch-factor } ]
```

詳細については、『プログラミングガイド』の「MULTI-FETCH 節」(Adabas) または『データベース管理システムインターフェイス』ドキュメントの「Natural for DB2」で「複数行の処理」(SQL)を参照してください。

## Starting/Ending 値の指定

開始値と終了値は、キーワード STARTING および ENDING (または、THRU) の後に、処理を開始／終了する値を示す定数またはユーザー定義変数を続けて指定できます。

指定された開始値が存在しない場合は、次に高い値が開始値として使用されます。高い値がまったくないときは、HISTOGRAM ループ内に入りません。

終了値を指定すると、終了値までの (終了値を含む) 値が読まれます。

フォーマットが A または B のディスクリプタに対しては、開始値および終了値として 16 進定数を指定できます。


構文オプション 1:

```
1 [ [STARTING] [ WITH
          FROM ] [VALUES] operand5 ] [ [ THRU
          ENDING AT ] operand6 ]
```

構文オプション 2:

```
2 [STARTING] [ WITH FROM ] [VALUES] operand5 TO operand6
```

構文オプション 3 :



**注意:** 図 3 の比較演算子を使用した場合は、オプション ENDING AT、THRU、および TO は使用できません。これらの比較演算子は READ ステートメントにも有効です。

オペランド定義テーブル :

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
operand5	C S	A U N P I F B D T L	可	不可
operand6	C S	A U N P I F B D T L	可	不可

構文要素の説明 :

<b>STARTING FROM ... ENDING AT/TO</b>	<p>STARTING FROM および ENDING AT 節は、ユーザー指定の値の範囲に読み込みを制限するために使用します。</p> <p>STARTING FROM 節 (=、EQ、EQUAL TO、または [STARTING] FROM) は、読み取り操作の開始値を決めます。開始値を指定すると、指定した値から読み取りが始まります。開始値が存在しない場合は、次に高い（または DESCENDING 読み込みでは低い）値が返されます。より高い（または DESCENDING ではより低い）値が存在しなければ、HISTOGRAM ループに入りません。</p> <p>値を終了値に制限するには、包括的な範囲を示す用語 THRU、ENDING AT、または TO を持つ ENDING AT 節を指定することができます。ディスクリプタフィールドが終了値を超えると、自動的なループ終了が実行されます。TO、THRU および ENDING AT キーワードの基本機能は、非常によく似ていますが、内部的に動作が異なります。</p>
<b>THRU/ENDING AT</b>	<p>THRU または ENDING AT が使用される場合、開始値だけがデータベースに供給され、データベースによって値が返された後に、Natural ランタイムシステムによって終了値のチェックが実行されます。</p>

	THRU および ENDING AT 節は、HISTOGRAM ステートメントをサポートするすべてのデータベースに対して使用することができます。
<b>TO</b>	キーワード TO を使用すると、開始値と終了値の両方がデータベースに送られ、Natural は値の範囲をチェックしません。終了値を超えると、データベースは「ファイルの終わり」に到達した場合と同様に対処し、データベースループを終了します。データベースにより完全な範囲チェックが行われるので、ASCENDING または DESCENDING のいずれの順序でブラウズしていても、常に範囲の小さい方の値が開始値になり、大きい方の値が終了値になります。

## 例

- [例 1 - HISTOGRAM ステートメント](#)
- [例 2 - 降順レコード読み取りによる HISTOGRAM ステートメント](#)
- [例 3 - 変数順序を使用した HISTOGRAM ステートメント](#)

### 例 1 - HISTOGRAM ステートメント

```

** Example 'HSTEX1S': HISTOGRAM (structured mode)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 CITY
END-DEFINE
*
LIMIT 8
HISTOGRAM EMPLOY-VIEW CITY STARTING FROM 'M'
  DISPLAY NOTITLE
    CITY 'NUMBER OF/PERSONS' *NUMBER *COUNTER
END-HISTOGRAM
*
END

```

プログラム **HSTEX1S** の出力：

CITY	NUMBER OF PERSONS	CNT
MADISON	3	1
MADRID	41	2
MAILLY LE CAMP	1	3
MAMERS	1	4
MANSFIELD	4	5
MARSEILLE	2	6
MATLOCK	1	7
MELBOURNE	2	8

レポートモードの例：[HSTEX1R](#)。

### 例 2 - 降順レコード読み取りによる HISTOGRAM ステートメント

```

** Example 'HSTDSCND': HISTOGRAM (with DESCENDING)
*****
DEFINE DATA LOCAL
1 EMPL VIEW OF EMPLOYEES
  2 NAME
END-DEFINE
*
HISTOGRAM (10) EMPL IN DESCENDING SEQUENCE FOR NAME FROM 'ZZZ'
  DISPLAY NAME *NUMBER
END-HISTOGRAM
END
    
```

プログラム **HSTDSCND** の出力：

```

Page          1                                05-01-13  13:41:03
      NAME                NMBR
-----
ZINN                      1
YOT                        1
YNCLAN                     1
YATES                      1
YALCIN                     1
YACKX-COLTEAU             1
XOLIN                      1
WYLLIS                     2
WULFRING                   1
WRIGHT                     1
    
```

### 例 3 - 変数順序を使用した HISTOGRAM ステートメント

```

** Example 'HSTVSEQ': HISTOGRAM (with VARIABLE SEQUENCE)
*****
DEFINE DATA LOCAL
1 EMPL VIEW OF EMPLOYEES
  2 NAME
*
1 #DIR      (A1)
1 #STARTVAL (A20)
END-DEFINE
*
SET KEY PF3 PF7 PF8
*
MOVE 'ADKINSON' TO #STARTVAL
*
    
```

```

HISTOGRAM (9) EMPL FOR NAME FROM #STARTVAL
  WRITE NAME *NUMBER
  IF *COUNTER = 5
    MOVE NAME TO #STARTVAL
  END-IF
END-HISTOGRAM
*
#DIR := 'A'
*
REPEAT
  HISTOGRAM EMPL IN VARIABLE #DIR SEQUENCE
    FOR NAME FROM #STARTVAL
    MOVE NAME TO #STARTVAL
    INPUT NO ERASE (IP=OFF AD=0)
      15/01 NAME *NUMBER
      // 'Direction:' #DIR
      // 'Press PF3 to stop'
      / ' PF7 to go step back'
      / ' PF8 to go step forward'
      / ' ENTER to continue in that direction'
    /*
    IF *PF-KEY = 'PF7' AND #DIR = 'A'
      MOVE 'D' TO #DIR
      ESCAPE BOTTOM
    END-IF
    IF *PF-KEY = 'PF8' AND #DIR = 'D'
      MOVE 'A' TO #DIR
      ESCAPE BOTTOM
    END-IF
    IF *PF-KEY = 'PF3'
      STOP
    END-IF
  END-HISTOGRAM
  /*
  IF *COUNTER(0250) = 0
    STOP
  END-IF
END-REPEAT
END

```

プログラム **HSTVSEQ** の出力：

```

Page      1                                05-01-13  13:50:31

ADKINSON                                8
AECKERLE                                1
AFANASSIEV                              2
AHL                                      1
AKROYD                                   1
ALEMAN                                   1
ALESTIA                                  1
ALEXANDER                                5

```

## HISTOGRAM

---

ALLEGRE 1

MORE

**Enter** キーを押した後：

Page 1 05-01-13 13:50:31

ADKINSON	8
AECKERLE	1
AFANASSIEV	2
AHL	1
AKROYD	1
ALEMAN	1
ALESTIA	1
ALEXANDER	5
ALLEGRE	1

AKROYD 1

Direction: A

Press PF3 to stop  
PF7 to go step back  
PF8 to go step forward  
ENTER to continue in that direction

# 71 IF

---

■ 機能 .....	478
■ 構文説明 .....	478
■ 例 .....	479

## ストラクチャードモード構文

```
IF logical-condition
  [THEN] statement
  [ELSE statement]
END-IF
```

## レポートモード構文

```
IF logical-condition
  [THEN] { statement
           DOstatement ... DOEND }
  [ ELSE { statement
           DOstatement ... DOEND } ]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[DECIDE FOR](#) | [DECIDE ON](#) | [IF SELECTION](#) | [ON ERROR](#)

関連機能グループ：「[論理条件の処理](#)」

## 機能

IF ステートメントは、論理条件に基づいて、ステートメントあるいはステートメント群の実行を制御します。



**注意:** 条件に合うときに何もしない場合は、THEN 節に IGNORE ステートメントを指定する必要があります。

## 構文説明

<b>IF</b> <i>logical-condition</i>	IF ステートメントに続いて、指定されたステートメント（群）が実行されるかどうかを判定する論理的な条件です。
---------------------------------------	--



	<p>例：</p> <pre>IF #A = #B IF LEAVE-TAKEN GT 30 IF #SALARY(1) * 1.15 GT 5000 IF SALARY (4) = 5000 THRU 6000 IF DEPT = 'A10' OR = 'A20' OR = 'A30'</pre> <p>詳細については、『プログラミングガイド』の「論理条件基準」を参照してください。</p>
<b>THEN statement</b>	THEN 節では、論理条件が真である場合に実行される <i>statement</i> (群) を指定します。
<b>ELSE statement</b>	ELSE 節では、論理条件が真でない場合に実行される <i>statement</i> (群) を指定します。
<b>END-IF</b>	IF ステートメントを終了するには、Natural の予約語 END-IF を使用する必要があります。

## 例

```
** Example 'IFEX1S': IF (structured mode)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
  2 SALARY (1)
  2 BIRTH
1 VEHIC-VIEW VIEW OF VEHICLES
  2 PERSONNEL-ID
  2 MAKE
*
1 #BIRTH (D)
END-DEFINE
*
MOVE EDITED '19450101' TO #BIRTH (EM=YYYYMMDD)
SUSPEND IDENTICAL SUPPRESS
LIMIT 20
*
FND. FIND EMPLOY-VIEW WITH CITY = 'FRANKFURT'
      SORTED BY NAME BIRTH
      IF SALARY (1) LT 40000
        WRITE NOTITLE '*****' NAME 30X 'SALARY LT 40000'
      ELSE
        IF BIRTH GT #BIRTH
          FIND VEHIC-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (FND.)
          DISPLAY (IS=ON)
```

```

NAME BIRTH (EM=YYYY-MM-DD)
SALARY (1) MAKE (AL=8)
END-FIND
END-IF
END-IF
END-FIND
END

```

プログラム **IFEX1S** の出力：

NAME	DATE OF BIRTH	ANNUAL SALARY	MAKE	
BAECKER	1956-01-05	74400	BMW	
***** BECKER				SALARY LT 40000
BLOEMER	1979-11-07	45200	FIAT	
FALTER	1954-05-23	70800	FORD	
***** FALTER				SALARY LT 40000
***** GROTHE				SALARY LT 40000
***** HEILBROCK				SALARY LT 40000
***** HESCHMANN				SALARY LT 40000
HUCH	1952-09-12	67200	MERCEDES	
***** KICKSTEIN				SALARY LT 40000
***** KLEENE				SALARY LT 40000
***** KRAMER				SALARY LT 40000

レポートモードの例：**IFEX1R**。

# 72 IF SELECTION

---

■ 機能 .....	482
■ 構文説明 .....	482
■ 例 .....	483

ストラクチャードモード構文

```
IF SELECTION [NOT UNIQUE [IN [FIELDS]]] operand1 ...
  [THEN] statement ...
  [THEN statement...]
END-IF
```

レポートモード構文

```
IF SELECTION [NOT UNIQUE [IN [FIELDS]]] operand1...
  [THEN] { statement
           DO statement... DOEND }
  [ ELSE { statement
           DO statement... DOEND } ]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：**DECIDE FOR** | **DECIDE ON** | **IF** | **ON ERROR**

関連機能グループ：「[論理条件の処理](#)」

## 機能

IF SELECTION ステートメントは、連続して指定された英数字フィールドのうち1つのフィールドにのみ値が含まれていることを確認するために使用します。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A	A U           L C	可	不可

構文要素の説明：

<i>operand1</i>	<p>選択フィールド：</p> <p><i>operand1</i> として、チェックされるフィールドを指定します</p> <p>属性制御変数（フォーマットC）を指定する場合は、そのステータスがMODIFIEDに変更されたかどうかを考慮されます。属性制御変数にステータスMODIFIEDが割り当てられているかどうかを確認するには、MODIFIED オプションを使用します。</p>
<b>THEN</b> <i>statement</i>	<p>実行するステートメント：</p> <p>キーワード THEN に続くステートメントは、次のどれかが満たされるときに実行されます。</p> <ul style="list-style-type: none"> <li>■ 指定された全フィールド（<i>operand1</i>）の値が空値。</li> <li>■ 指定フィールドのうち複数フィールド（<i>operand1</i>）が空値ではない。</li> </ul> <p>一般的に、このステートメントは、INPUT ステートメントで表示されたマップに対して、端末ユーザーが1つの値だけを入力したことを確かめるために使用されます。</p> <p><b>注意:</b> 条件に合うときに何もしない場合は、THEN 節に IGNORE ステートメントを指定します。</p>
<b>END-IF</b>	<p>Natural 予約語 END-IF を使用して、IF SELECTION ステートメントを終了させる必要があります。</p>

## 例

```

** Example 'IFSEL': IF SELECTION
*****
DEFINE DATA LOCAL
1 #A (A1)
1 #B (A1)
END-DEFINE
*
INPUT 'Select one function:' //
    9X 'Function A:' #A
    9X 'Function B:' #B
*
IF SELECTION NOT UNIQUE #A #B
    REINPUT 'Please enter one function only.'
END-IF
*
IF #A NE ' '
    WRITE 'Function A selected.'
END-IF
IF #B NE ' '
    WRITE 'Function B selected.'

```

## IF SELECTION

---

```
END-IF  
*  
END
```

プログラム **IFSEL** の出力：

```
Select one function:
```

```
Function A:
```

```
Function B:
```

機能 **A** を選択して確認した後：

```
Page      1
```

```
05-01-17 11:04:07
```

```
Function A selected.
```

# 73 IGNORE

---

▪ 機能 .....	486
▪ 例 .....	486

IGNORE

## 機能

---

IGNORE ステートメントは、「空」のステートメントであり、それ自体は何の機能も実行しません。

アプリケーションの開発段階で、1つ以上のステートメントが必要であるが、後からコーディングするつもりでいるようなステートメントブロック内（例：[AT BREAK](#) や [AT START OF DATA/AT END OF DATA](#) 内）に、一時的に IGNORE を挿入することができます。これにより、不完全なステートメントブロックのままでも、エラーを起こすことなくアプリケーションの他の部分を開発できます。

条件に合うときには、どの機能も実行しない場合、[IF](#) または [DECIDE FOR](#) と同様に、条件ステートメントに IGNORE ステートメントを使用する必要があります。

## 例

---

```
...
...
AT TOP OF PAGE
  IGNORE      /* top-of-page processing still to be coded
END-TOPPAGE
...
...
```



# 74 INCLUDE

---

■ 機能 .....	488
■ 構文説明 .....	488
■ 例 .....	489


```
INCLUDE copycode-name [operand1 ... 99]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

INCLUDE ステートメントは、コピーコードタイプの外部オブジェクトのソース行を別のオブジェクトにコンパイル時に取り込みます。

INCLUDE ステートメントはコンパイル時に評価されます。コピーコードのソース行は、INCLUDE ステートメントを含むプログラムのソース行に、物理的に組み込まれるわけではありません。プログラムのコンパイル時に組み込まれた結果が、オブジェクトモジュールとなります。

 **注意:** INCLUDE ステートメントを含むソースコード行に、他のステートメントは指定できません。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C	A U	不可	不可

構文要素の説明：

<i>copycode-name</i>	<p><i>copycode-name</i> には、挿入するソースを持つコピーコードの名前を指定します。</p> <p><i>copycode-name</i> にはアンパサンド (&amp;) を含むことができます。この文字はコンパイル時に <b>Natural</b> システム変数 *LANGUAGE の現在の値で置き換えられます。この機能で複数言語のコピーコード名を使用できます。</p> <p>指定するオブジェクトは、コピーコードタイプにする必要があります。コピーコードは、INCLUDE ステートメントを持つプログラムと同一ライブラリ、または STEPLIB (デフォルトの STEPLIB は SYSTEM) に保存します。</p> <p>コピーコードのソースが変更された場合、その変更をオブジェクトコードに反映させるには、変更したコピーコードを使用する全プログラムを再コンパイルする必要があります。</p> <p>コピーコードのソースコードは構文的に完全なステートメントで構成されている必要があります。</p>
----------------------	--

<i>operand1</i>	<p>組み込むコピーコードの値をダイナミックに挿入できます。この値は <i>operand1</i> で指定します。</p> <p>コピーコードでは、値は次の表記で参照されます。</p> <p><code>&amp;n&amp;</code></p> <p>つまり、値が挿入される位置を <code>&amp;n&amp;</code> で指定します。 <i>n</i> は INCLUDE ステートメントで渡される各値の順番号です。例えば、<code>&amp;3&amp;</code> 指定は、ステートメントで指定された 3 番目の値を参照します。</p> <p>コピーコード内の各 <code>&amp;n&amp;</code> に対して、INCLUDE ステートメントに値を指定する必要があります。例えば、コピーコードが <code>&amp;5&amp;</code> を含んでいる場合、<i>operand1</i> は少なくとも 5 回指定する必要があります。</p> <p>1つのコピーコードパラメータ (<code>&amp;n&amp;</code>) を、空白を入れずに別のコピーコードパラメータの後に指定することができます (<code>&amp;1&amp;&amp;2&amp;&amp;3&amp;</code>)。この方法は、複数のコピーコードパラメータをソースへ連結するために使用されます。</p> <p>文字列を、空白を入れずに1つまたは複数のコピーコードパラメータに続けることができます (<code>&amp;1&amp;abc</code> または <code>&amp;1&amp;&amp;2&amp;abc</code>)。この方法は、複数のコピーコードパラメータに文字列を連結するために使用されます。</p> <p><b>注意:</b> <code>&amp;n&amp;</code> は識別子の有効な部分であるため、この表記法は、上記の他の位置でコピーコードパラメータ代用 (<code>abc&amp;1&amp;</code> または <code>&amp;1&amp;abc&amp;2&amp;</code>) として使用することはできません。つまり、文字列は、コピーコードパラメータの後にのみ指定可能で、前または間に指定することはできません。</p> <p>INCLUDE ステートメントに値を指定してもコピーコード内で参照しなければ無視されます。</p>
-----------------	--

## 例

- 例 1 - コピーコードを含む INCLUDE ステートメント
- 例 2 - パラメータのあるコピーコードを含む INCLUDE ステートメント
- 例 3 - ネスト構造のコピーコードを使用した INCLUDE ステートメント

## ■ 例4-コピーコードに連結パラメータを含む INCLUDE ステートメント

### 例1-コピーコードを含む INCLUDE ステートメント

INCLUDE ステートメントを含むプログラム：

```
** Example 'INCEX1': INCLUDE (include copycode)
*****
*
WRITE 'Before copycode'
*
INCLUDE INCEX1C
*
WRITE 'After copycode'
*
END
```

含めるコピーコード **INCEX1C**：

```
** Example 'INCEX1C': INCLUDE (copycode used by INCEX1)
*****
*
WRITE 'Inside copycode'
```

プログラム **INCEX1** の出力：

```
Page      1                                05-01-25  16:26:36
Before copycode
Inside copycode
After copycode
```

### 例2-パラメータのあるコピーコードを含む INCLUDE ステートメント

INCLUDE ステートメントを含むプログラム **INCEX2**：

```
** Example 'INCEX2': INCLUDE (include copycode with parameters)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
END-DEFINE
*
*
INCLUDE INCEX2C 'EMPL-VIEW' 'NAME' '''ARCHER''' '20' '''BAILLET'''
*
END
```

含めるコピーコード **INCEX2C** :

```

** Example 'INCEX2C': INCLUDE (copycode used by INCEX2)
*****
* Transferred parameters from INCEX2:
*
* &1& : EMPL-VIEW
* &2& : NAME
* &3& : 'ARCHER'
* &4& : 20
* &5& : 'BAILLET'
*
*
READ (&4&) &1& BY &2& = &3&
  DISPLAY &2&
  IF &2& = &5&
    WRITE 5X 'LAST RECORD FOUND' &2&
    STOP
  END-IF
END-READ
*
* Statements above will be completed to:
*
* READ (20) EMPL-VIEW BY NAME = 'ARCHER'
*   DISPLAY NAME
*   IF NAME = 'BAILLET'
*     WRITE 5X 'LAST RECORD FOUND' NAME
*     STOP
*   END-IF
* END-READ

```

プログラム **INCEX2** の出力 :

```

Page      1                                05-01-25  16:30:43
      NAME
-----
ARCHER
ARCONADA
ARCONADA
ARNOLD
ASTIER
ATHERTON
ATHERTON
ATHERTON
AUBERT
BACHMANN
BAECKER
BAECKER
BAGAZJA

```

```
BAILLET
  LAST RECORD FOUND BAILLET
```

### 例3 - ネスト構造のコピーコードを使用した INCLUDE ステートメント

INCLUDE ステートメントを含むプログラム：

```
** Example 'INCEX3': INCLUDE (using nested copycodes)
*****
DEFINE DATA LOCAL
1 #A (I4)
END-DEFINE
*
MOVE 123 TO #A
WRITE 'Program INCEX3 ' '=' #A
*
INCLUDE INCEX31C '#A' '5'          /* source line is #A := 5
*
*
MOVE 300 TO #A
WRITE 'Program INCEX3 ' '=' #A
*
INCLUDE INCEX32C '''#A'''  '''20''' /* source line is #A := 20
*
WRITE 'Program INCEX3 ' '=' #A
END
```

含めるコピーコード INCEX31C：

```
** Example 'INCEX31C': INCLUDE (copycode used by INCEX3)
*****
* Transferred parameters from INCEX3:
*
* &1& : #A
* &2& : 5
*
*
&1& := &2&
*
WRITE 'Copycode INCEX31C' '=' &1&
```

含めるコピーコード **INCEX32C** :

```
** Example 'INCEX32C': INCLUDE (copycode used by INCEX3)
*****
* Transferred parameters from INCEX3:
*
* &1& : '#A'
* &2& : '20'
*
*
WRITE 'Copycode INCEX32C' &1& &2&
*
INCLUDE INCEX31C &1& &2&
```

プログラム **INCEX3** の出力 :

```
Page          1                                05-01-25  16:35:36

Program INCEX3   #A:          123
Copycode INCEX31C #A:           5
Program INCEX3   #A:          300
Copycode INCEX32C #A 20
Copycode INCEX31C #A:           20
Program INCEX3   #A:           20
```

#### 例 4 - コピーコードに連結パラメータを含む INCLUDE ステートメント

**INCLUDE** ステートメントを含むプログラム :

```
** Example 'INCEX4': INCLUDE (with concatenated parameters in copycode)
*****
DEFINE DATA LOCAL
1 #GROUP
  2 ABC(A10) INIT <'1234567890'>
END-DEFINE
*
INCLUDE INCEX4C '#GROUP.' 'ABC' 'AB'
*
END
```

含めるコピーコード **INCEX4C** :

```
** Example 'INCEX4C': INCLUDE (copycode used by INCEX4)
*****
* Transferred parameters from INCEX4:
*
* &1& : #GROUP.
* &2& : ABC
* &3& : AB
*
*
WRITE  '=' &2&           /* 'ABC'           results into ABC
WRITE  '=' &1&ABC        /* '#GROUP.' ABC   results into #GROUP.ABC
WRITE  '=' &1&&2&        /* '#GROUP.' 'ABC' results into #GROUP.ABC
WRITE  '=' &1&&3&C       /* '#GROUP.' 'AB' C results into #GROUP.ABC
```

プログラム **INCEX4** の出力 :

```
Page      1                                     05-01-25  16:37:59
ABC: 1234567890
ABC: 1234567890
ABC: 1234567890
ABC: 1234567890
```



# 75 INPUT

---

■ 機能 .....	496
■ 入力モード .....	496
■ データ入力 .....	498
■ SB - 選択ボックス .....	500
■ エラー修正 .....	500
■ 画面分割 .....	501

構文については別途説明します。以下を参照してください。

- **INPUT 構文 1** - ダイナミック画面レイアウトの指定
- **INPUT 構文 2** - 定義済みマップレイアウトの使用

『プログラミングガイド』の「画面設計」／「ウィンドウ」も参照してください。

関連ステートメント：**DEFINE WINDOW** | **REINPUT** | **SET WINDOW**

関連機能グループ：「[対話型処理用の画面生成](#)」

## 機能

---

INPUT ステートメントは、会話型モードの場合、データ入力のための形式化された画面やマップを作成するために使用します。

Natural スタック (**STACK** ステートメント参照) と結合して使用することもできます。また、メインフレーム環境では、バッチモードで実行しているプログラムに対するユーザーデータを提供することもできます。

Natural RPC については、『*Natural* リモートプロシージャコール (RPC)』ドキュメントの「サーバーに対する *Natural* ステートメントの注意事項」を参照してください。

## 入力モード

---

INPUT ステートメントは、スクリーン、フォームまたはキーワード／デリミタの各モードで使用できます。スクリーンモードは、通常ビデオ端末で使用します。フォームモードは、TTY 端末で使用します。デリミタモードは、TTY 端末およびバッチモードで使用します（メインフレームコンピュータの場合）。デフォルトはスクリーンモードです。

セッションパラメータ IM または端末コマンド %F および %D を使用して、入力モードを変更できます。

## スクリーンモード

スクリーンモードでは、INPUTステートメントを実行すると、フィールドや位置指定に従って画面が表示されます。画面のメッセージ行は、Naturalのエラーメッセージに使用されます。メッセージ行の位置（画面の最上段または下段）は、端末コマンド%Mを使用して変更できます。端末ユーザーは、Tabキーで個々のフィールドに位置付けることができます。

Naturalには画面ウィンドウ処理が用意されています。そのため、論理画面マップのレイアウトは物理画面サイズより大きく作成できます（理論上250桁×250行、ただし内部画面バッファにより制限されます）。

ウィンドウ端末コマンドの%Wで、論理的および物理的ウィンドウの位置や大きさを変更できます（ウィンドウ制御の詳細は、端末コマンド%W参照）。

物理画面に入り切らない入力フィールド（AD=AまたはAD=M）には、次の規則が適用されます。

- フィールドの始まりがウィンドウ内でない入力フィールドは常に保護されます。
- ウィンドウ内で始まりウィンドウ外で終わる入力フィールドは、フィールドに含まれる値がウィンドウ内に完全に表示できない場合にのみ保護されます。この場合、フィールド長ではなく値の長さがウィンドウのサイズを超えているかどうかを決定します。充填文字（プロファイルパラメータFCまたはセッションパラメータADで指定）は値の一部とはみなされません。
- このように保護された入力フィールドをアクセスまたは処理するには、ウィンドウサイズを調節してフィールドや値の全体を画面表示できるようにする必要があります（端末コマンド%Wを参照）。

## 非スクリーンモード

INPUTステートメントは、行単位デバイスのオペレーション、またはシーケンシャルファイルからのバッチ入力処理に使用することもできます。

スクリーンモードで定義した同じマップレイアウトを、非スクリーンモードで使用することもできます。

フォームモードおよびキーワード/デリミタモードは、行モードの画面レイアウトと同様に、またはマップレイアウトなしでデータだけを処理することによって入力を処理できます。

以下の項目も参照してください。

- [非スクリーンモードでのINPUTステートメントの使用](#)
- [バッチモードでのINPUTステートメントの使用](#)
- [Naturalスタックデータの処理](#)

## データ入力

英数字フィールドのデータは左詰めで入力する必要があります。空白も含めてすべての文字に意味があります。データは1文字ごとにバイト単位で内部フィールドに割り当てられます。英数字フィールドに入力されるデータは、入力の際はチェックされません。

小文字および大文字の変換はパラメータ AD=T と AD=W、または端末コマンド %L と %U で制御できます。

数字フィールドのデータは、入力フィールドの中のどこでも入力できます。リーディングおよび（または）トレーリングブランク、リーディングゼロ、リーディングサインおよび1つの小数点を使用できます。Natural は、フィールドの内部定義に従って、値の桁合わせをします。SG=OFF が指定されている場合は、Natural は符号桁を確保しません。P フォーマットで定義されたフィールドのデータは、10進数形式で入力する必要があります。必要に応じて10進がパック形式に変換されます。全部が空白のフィールドは、値0とみなされます。数字フィールドのデータは、リーディングおよび（または）トレーリングブランク、オプションで1つのリーディングサイン、オプションで1つの小数点、および数字以外のものが値に含まれていないかどうかチェックされます。小数点が入力されていないければ、値の右端にあると仮定します。

バイナリフィールドに対するデータは、入力フィールドのすべての桁に対して入力する必要があります（1バイトに対して2桁が対応します）。16進数文字（0~9、A~F）だけが使用できます。空白文字（ASCII では H'20'、または EBCDIC では H'40'）は有効で、バイナリゼロに変換されます。バイナリフィールドのデータは、16進文字かどうかチェックされます。

フォーマット L のフィールドに対するデータには、空白（偽）または空白以外（真）を入力できます。

フォーマット F、D および T に対するデータは、F、D および T の定数の定められた規約に従って入力します。

### 数値編集マスクフリーモード

フィールド要素内に、フィールド内容の表現を編集マスクでフォーマットすることができます。編集マスクは次の2つの目的に使用されます。

- 画面にフィールド表示するためのレイアウトの構築。
- 文字列が修正され、Enter が押されたときに、入力された文字列からフィールドデータを抽出するため。

入力される新しいデータ値は編集マスクのフォーマットと完全に一致している必要があるため、追加の挿入文字付きで表示されるフィールドデータのフォーマットを改善する利点は実際には不都合である可能性があります。

例：

```
SET GLOBALS ID=; DC=,
RESET N (N7,3)
INPUT N (AD=M EM=Z'.'ZZZ'.'ZZZ,999EUR)
END
```

出力値	表示	入力値	必要な入力	エラーになる入力
0	,000EUR	1	1,000EUR	1 1EUR 01,000EUR
1234	1.234,000EUR	1234567	1.234.567,000EUR	1234567 1.234.567 1.234.567EUR
0,123	,123EUR	1,234	1,234EUR	1,234

編集マスクを持つ数値フィールドを入力するためのもう1つのオプションは、編集マスクフリーモードと呼ばれる代替 INPUT モードを使用することです。（プロファイルパラメータ EMFM 指定のセッション起動時、または端末コマンド %FM+ により Natural セッション実行中に）アクティブにされると、編集マスク挿入文字のすべてまたは一部が入力から除外されることがあります。

ただし、挿入文字の連続文字列が編集マスクに（以下の例の EUR のように）表示されると、文字列を完全に供給するか、または除外することのみが可能となります。供給されるオプションまたは必須の桁数（編集マスク文字 Z および 9）には影響しません。

例：編集マスクフリーモードを有効にする

```
SET GLOBALS ID=; DC=,
SET CONTROL 'FM+' /* activate numeric Edit Mask Free Mode
RESET N (N7,3)
INPUT N (AD=M EM=Z'.'ZZZ'.'ZZZ,999EUR)
END
```

入力値	可能な入力	エラーになる入力
1	1 1,0 001 1,00EUR 0.001 1,EUR	1EUR
1234567	1234567 1.234.567 1234.567 1234567,0 1.234.567,0 1.234.567,EUR	1.234.567EUR

入力値	可能な入力	エラーになる入力
	1.234.567,0EUR 1.234.567,000EUR	
1,234	1,234 1,234EUR 001,234 0.001,234EUR 00001,234EUR	1,234EU



**注意:** 編集マスクフリーモードは INPUT にのみ適用され、**MOVE EDITED** ステートメントでは無視されます。

## SB - 選択ボックス

INPUT ステートメントの選択ボックスは、メインフレームコンピュータでのみ利用可能です。Windows では、選択ボックスはマップエディタでのみ定義できます。UNIX および OpenVMS では、選択ボックスは定義できず、Windows またはメインフレーム環境からインポートされた場合には無視されます。

選択ボックスは入力フィールドに付加することができます。選択ボックスはコーディングをプログラム内で直接行うことができるため、フィールドにヘルプルーチンを添付する方法に代わる最適な選択肢です。ヘルプルーチンと同様に、特別なプログラムは必要ありません。

詳細については、『パラメータリファレンス』でセッションパラメータ SB を参照してください。

## エラー修正

INPUT フィールドに入力した値がフォーマットやフィールドの編集マスクと一致しない場合、（プログラムの実行を終了せずに）エラーメッセージが表示され、エラーのあるフィールドにカーソルが位置付けられます。そこでユーザーは正しい値を入力でき、その後処理は継続されます。

---

## 画面分割

---

一般的に、各 INPUT ステートメントは、新しいページ（または端末画面）を生成します。AT **END OF PAGE** ステートメント内に指定された INPUT ステートメントは、新しい画面を生成しません。この機能を使用することにより分割画面が作成できます。分割画面の上部には複数の行を表示し、画面の下部は会話のため、入力マップの作成に使用することができます。PS パラメータ（ページサイズ）を、**SET GLOBALS** または **FORMAT** ステートメントで指定し、同一物理画面に入力マップが作成できるように論理ページサイズを設定します。

最初の INPUT 行は最終表示行の後に位置付けられます。ただし **NO ERASE** オプションを使用した場合、最初の INPUT 行はページの先頭に位置付けられます。






# 76 INPUT 構文 1 - ダイナミック画面レイアウトの指定

---

- INPUT 構文 1 - 説明 ..... 504
- 例 - 構文 1 ..... 513

## INPUT 構文 1- ダイナミック画面レイアウトの指定

この形式の INPUT ステートメントは、INPUT 画面のレイアウトの作成、またはシーケンシャル入力ファイルからバッチモードで（メインフレームコンピュータで）読み取る INPUT データレイアウトの作成に使用されます。



構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## INPUT 構文 1 - 説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A G N A U N P I F B D T L G		可	可

構文要素の説明：

<b>INPUT</b> <b>WINDOW='window-name'</b>	<p>オプションWINDOW='window-name'では、INPUTステートメントを指定されたウィンドウに対して実行することを指定します。指定されたウィンドウはDEFINE WINDOWステートメントで定義されている必要があります。指定されたウィンドウは、INPUTステートメントの期間中のみアクティブになり、INPUTステートメントが実行されると自動的に非アクティブになります。</p> <p>ステートメント「<a href="#">DEFINE WINDOW</a>」および「<a href="#">SET WINDOW</a>」も参照してください。</p>
<b>NO ERASE</b>	<p>NO ERASEを指定すると、INPUTステートメントの画面マップが既存の画面の上にオーバーレイされます。画面の内容は消去されません。</p> <p>ここでの画面とは、物理画面ではなく論理画面を意味します。</p>

	<p>画面に存在する保護されていないすべてのフィールドは、保護（表示専用）フィールドに変換されます。古いデータは、新しいレイアウトが表示されるまで画面に残ります。新しい画面の内容のフィールドが既存のフィールドを部分的にオーバーレイする場合は、新しいフィールドの前の1文字と、既存のフィールド内の次の文字が空白で置換されます。</p>
<p><i>statement-parameters</i></p>	<p>1つ以上のパラメータを、カッコで囲んで INPUT ステートメントまたは表示項目の直後に指定できます。</p> <p>INPUT ステートメントで指定できるパラメータのリストについては、下記の「ステートメントのパラメータ」を参照してください。</p> <p>この方法で指定した各パラメータは、その前に指定された GLOBALS コマンド、SET GLOBALS または FORMAT ステートメントのパラメータを上書きします。複数のパラメータを指定する場合は、各エントリ間に1つ以上の空白を配置する必要があります。エントリを2行のステートメント行に分割することはできません。</p> <p>ここで適用されるパラメータ設定は、変数フィールドにのみ関連し、テキスト定数には影響しません。テキスト定数にフィールド属性を設定する場合は、この要素に属性を明示的に設定する必要があります。</p> <p>例：</p> <pre> DEFINE DATA LOCAL 1 VARI (A4)      INIT &lt;'1234'&gt;          /*   Displays END-DEFINE      /*   as FORMAT AD=M     /*   ----- INPUT           'Text'                 VARI          /*   Text 1234 INPUT (PM=I)   'Text'                 VARI          /*   Text 4321 INPUT          'Text' (PM=I)          VARI (PM=I)   /*   txeT 4321 INPUT          'Text' (PM=I)          VARI          /*   txeT 1234 END                 </pre> <p>ステートメントレベルおよび要素レベルでのパラメータの使用例を次に示します。</p>
<p><i>WITH TEXT-option</i></p>	<p>WITH TEXT は、メッセージ行に表示されるテキストを提供するために使用します（<a href="#">下記</a>の対応するセクションを参照）。</p>
<p><i>MARK-option</i></p>	<p>MARK POSITION を使用して、フィールド内の特定の位置にカーソルを配置できます。下記の「<a href="#">MARK オプション</a>」を参照してください。</p>
<p><i>ALARM-option</i></p>	<p>このオプションは、INPUT ステートメントの実行時にアクティブ化される端末の音声アラーム機能を提供します。下記の「<a href="#">Alarm オプション</a>」を参照してください。</p>
<p>その他の構文要素 (<i>nX</i>、<i>nT</i>、<i>x/y</i>、<i>operand1</i> など)</p>	<p>下記の「<a href="#">フィールドの位置、文字列、属性の指定</a>」を参照してください。</p>

## ステートメントのパラメータ

INPUT ステートメントで指定可能なパラメータ		指定 (S=ステートメントレベル、E=要素レベル)
AD	属性定義	SE
AL	出力の英数字長	SE
BX	ボックス定義	SE
CD	カラー定義	SE
CV	制御変数	SE
DF	日付フォーマット	SE
DL	出力の表示長	SE
DY	ダイナミック属性	SE
EM	編集マスク	SE
FL	浮動小数点仮数長	SE
HE	ヘルプルーチン	SE
IP	入力プロンプトのテキスト	SE
LS	行サイズ	S
MC	マルチプルバリューフィールド数	S
MS	手動による省略	S
NL	出力の数値長	SE
PC	ピリオディックグループ数	S
PM	出力モード	SE
PS	ページサイズ*	S
SB	選択ボックス	E
SG	符号の位置	SE
ZP	ゼロ出力	SE

\* 配列のオカレンス数が PS 値を超えた場合は、NAT0303 エラーが出力されます。

各セッションパラメータの詳細については、『パラメータリファレンス』を参照してください。

## WITH TEXT オプション

```
[WITH] TEXT { * operand1
               operand2 } [(attributes)][,operand3] ... 7
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
operand1	C S	N P I B*	可	可
operand2	C S	A	可	可
operand3	C S	A N P I F B D T L	可	可

\*operand1 のフォーマット B は、4 以下の長さでのみ使用できます。

WITH TEXT は、メッセージ行に表示されるテキストを提供するために使用します。これは通常、画面処理やエラー修正のために実行する必要がある操作を示すメッセージです。


構文要素の説明：

operand1	<p><b>Natural</b> メッセージファイルからのメッセージテキスト：</p> <p>operand1 は、Natural メッセージファイルから取得するメッセージテキストの番号を表します。</p> <p>ユーザー定義メッセージまたは Natural システムメッセージのいずれかを取得できます。</p> <ul style="list-style-type: none"> <li>■ 最大4桁の正の値（954 など）を指定すると、ユーザー定義メッセージを取得できます。</li> <li>■ 最大4桁の負の値（-954 など）を指定すると、Natural システムメッセージを取得できます。</li> </ul> <p>REINPUT ステートメントの説明にある「<a href="#">例4-WITH TEXT オプション</a>」も参照してください。</p> <p>Natural メッセージファイルは、関連ドキュメントで説明しているように、SYSERR ユーティリティで作成およびメンテナンスします。</p>
operand2	<p>メッセージテキスト：</p> <p>operand2 は、メッセージ行に配置されるメッセージを表します。</p> <p>REINPUT ステートメントの説明にある「<a href="#">例4-WITH TEXT オプション</a>」も参照してください。</p>
attributes	<p>operand1/2 にはさまざまな出力属性を割り当てることができます。これらの属性と構文については、以下の「<a href="#">出力属性</a>」で説明します。</p>
operand3	<p>メッセージテキストのダイナミック置換：</p> <p>Operand3 は、英数字、テキスト定数、または変数名を表します。</p>

	<p>提供された値は、<i>operand1</i> または <i>operand2</i> のいずれかで指定されるメッセージテキストの一部を置き換えるために使用されます。</p> <p>:<i>n</i>: は <i>operand3</i> の内容を参照するためにメッセージテキスト内に指定します。<i>n</i> には <i>operand3</i> のオカレンス (1~7) を指定します。</p> <p>REINPUT ステートメントの説明にある「<a href="#">例4-WITH TEXT オプション</a>」も参照してください。</p> <p><b>注意:</b> 複数の <i>operand3</i> を指定する場合は、それぞれをコンマで区切る必要があります。コンマを小数点文字として (セッションパラメータ DC で定義) 使用し、数値定数を <i>operand3</i> として指定している場合は、コンマの前後に空白を挿入して、コンマが小数点文字と解釈されないようにします。または、複数の <i>operand3</i> を INPUT 区切り文字 (セッションパラメータ ID で指定) で区切ることができます。ただし、INPUT ステートメント構文ではスラッシュは別の意味を持つため、ID=/ (スラッシュ) の場合は使用できません。</p> <p>先頭のゼロまたは末尾の空白は、メッセージに表示される前にフィールド値から削除されます。</p>
--	---

出力属性

*attributes* は、テキスト表示に使用される出力属性を示します。可能な属性：



指定可能なセッションパラメータ値については、『パラメータリファレンス』ドキュメントの該当するセクションを参照してください。

- 「AD - 属性定義」の「フィールド表現」
- CD - カラー定義
- BX - 外枠の定義
- PM - 出力モード

**注意:** コンパイラは、実際には1つの出力フィールドに複数の属性値を受け入れます。例えば、「AD=BDI」と指定できます。ただし、この場合は最後の値のみが適用されます。示した例では、値 I のみが有効になり、出力フィールドは強調表示されます。

## MARK オプション

MARK POSITION を使用して、フィールド内の特定の位置にカーソルを配置できます。

```
MARK [POSITION operand4 [IN]] [FIELD] { operand1
                                         *fieldname }
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand4</i>	C S	N P I	可	可
<i>operand1</i>	C S A	N P I	可	可

構文要素の説明：

<b>MARK *<i>fieldname</i></b>	<i>*fieldname</i> は、フィールド名を使用して位置付けるために使用します。
<b>MARK POSITION</b> <i>operand4</i>	<p>MARK POSITION では、<i>operand1</i> で指定したフィールド内の <i>operand4</i> で指定した特定の位置にカーソルを置くことができます。 <i>operand4</i> に小数桁を含めることはできません。</p> <p>例：</p> <pre>MARK 3 MARK #A MARK *# MARK POSITION 3 IN #A</pre>
<i>operand1</i>	<p><i>operand1</i> では、マークするフィールドを指定します。</p> <p>INPUT ステートメントで指定された各フィールド属性 AD=Aまたは AD=M（保護されていないフィールド）には、1 から開始するフィールド参照番号が割り当てられます。 MARK オプションは、指定されたフィールド番号にカーソルを配置します。</p>

## ALARM オプション

このオプションは、INPUT ステートメントの実行時にアクティブ化される端末の音声アラーム機能を提供します。この機能を使用するには、対応するハードウェアが使用可能である必要があります。

```
[[AND] [SOUND] ALARM]
```

デフォルトのプロンプトテキスト

セッションパラメータ IP (入力プロンプト) が IP=OFF に設定されていない限り、INPUT ステートメントで使用されるフィールドのフィールド名が、フィールド値の前に (フォームモード)、またはフィールドを選択するためのプロンプトキーワードとして (キーワード/デリミタモード) 表示されます。このデフォルトのフィールド名は、'text' 要素 (デフォルト名を置換) または '-' (デフォルトフィールド名の表示を抑制) をフィールド名の直前に指定することで上書きできます。

## フィールドの位置、文字列、属性の指定

$\begin{bmatrix} nX \\ nT \\ x/y \end{bmatrix}$	'text' [(attributes)]	$\begin{bmatrix} *IN \\ *OUT \\ *OUTIN \end{bmatrix} \begin{matrix} \{operand1 \\ [(parameter(s))]\} \end{matrix}$
	'c'(n) [(attributes)]	
	'-'	
	'='	
	/ ...	

フィールドの位置、属性の割り当て、およびテキストの作成には複数の表記を使用できます。

nX	n 個のスペースがフィールド間に挿入されます。 <b>注意:</b> この表記により、列の間に n 個のスペースが挿入されます。n をゼロにすることはできません。						
nT	位置 n を出力するように位置指定 (タブ設定) されます。						
x/y	次の要素を行 x の列 y から開始して配置します。y をゼロにすることはできません。同じ行で後方に位置指定することはできません。						
'text'	text が書き込み保護状態で表示されます。「テキスト表記」の「ステートメントで使用するテキストの定義」も参照してください。						
'c'(n)	'text' と同一ですが、文字 c が n 回表示されます。n は 1~132 にする必要があります。「テキスト表記」の「フィールド値の前に n 回表示する文字の定義」も参照してください。						
attributes	表示に使用される属性を示します。下記の「属性」を参照してください。						




`-`	<p>マイナス記号：</p> <p>フィールドの前に配置された場合、`-` はプロンプトテキストとしてのフィールド名の生成を抑制します。</p> <p><b>注意:</b> フィールドの前のテキスト文字列は、フィールド名をプロンプトテキストとして置換します。</p>	
`=`	<p>等号：</p> <p>フィールドの前に `=` を配置した場合は、後にフィールドの内容が続くフィールドヘッダーが表示されます。</p>	
`/`	<p>スラッシュ記号：</p> <p>フィールドまたはテキスト要素の間に `/` を配置した場合は、次の出力行の先頭に位置指定されます。</p> <p>フィールドの内容は、属性設定 AD=A、AD=0 および AD=M を使用して、それぞれ入力、出力のみ、変更用の出力に指定できます。デフォルトは AD=A です。AD=A（入力のみ）または AD=M（変更用の出力）が指定されたすべてのフィールドでは、保護されていないフィールドが画面に作成されます。このようなフィールドの値はユーザーが入力できます。TTY デバイスの場合、変更用の出力フィールドは、新しい値を入力できるように、フィールドのサイズの 2 倍（出力用に 1 つと入力用に 1 つ）を占めます。表示不能として指定された入力フィールド（AD=A/M）は、TTY デバイスで常に新しい行で開始します。</p> <p>例：</p> <pre>INPUT #A (AD=A) #B (AD=0) #C (AD=M)</pre> <p>#A は、保護されていない入力フィールドです。つまり、フィールドに値が入力されます。</p> <p>#B は、書き込み保護で表示されるフィールドです。つまり、フィールドには値を入力できません。</p> <p>#C は、現在の値が表示されるフィールドであり、フィールドに新しい値を入力することで値を変更できます。</p>	
`*IN`、`*OUT` および `*OUTIN`	<p>それぞれ属性 AD=A、AD=0、AD=M と同等です。</p> <p><b>注意:</b> 変更不能なシステム変数が INPUT ステートメントで使用されている場合は、値が出力専用フィールド AD=0 または `*OUT` 属性として表示されます。</p>	

<p><i>operand1</i></p>	<p>フィールド指定：</p> <p><i>operand1</i> は、使用するフィールドを表します。データベースフィールドまたはユーザー定義変数を指定できます。</p> <p>Natural は、データエリアの各フィールドの内容を INPUT ステートメントにマップするため、移動操作は不要です。</p> <p>データベースフィールドの内容が INPUT 処理の結果として変更される場合は、データエリアにある値のみが変更されます。適切なデータベースステートメント <b>UPDATE/STORE</b> を使用して、データベースの内容を変更する必要があります。</p> <p>データベースフィールドのグループの名前が INPUT ステートメントで参照されている場合は、そのグループに属するすべてのフィールドが入力フィールドとして個別に使用されます。</p> <p>配列内のオカレンスの範囲に対する参照が行われた場合は、すべてのオカレンスが個々に入力フィールドとして処理されますが、プロンプトテキストは個々のオカレンスに対しては作成されず、最初のオカレンスに対してのみ作成されます。</p> <p>メインフレームコンピュータでは、実行時にオカレンス数を変更できる、範囲による配列は指定できません。</p>	
<p><i>parameter(s)</i></p>	<p>1つまたは複数のパラメータをカッコで囲んで <i>operand1</i> の直後に指定できます (下記の表と例を参照)。</p> <p>指定した各パラメータは、その前に指定された GLOBALS コマンド、<b>SET GLOBALS</b> (レポートモード) または <b>FORMAT</b> ステートメントのパラメータを上書きします。複数のパラメータを指定する場合は、個々のパラメータを1つ以上の空白で区切る必要があります。各パラメータ指定を2行のステートメント行に分割することはできません。</p> <p>ここで適用されるパラメータ設定は、変数フィールドにのみ関連し、テキスト定数には影響しません。テキスト定数にフィールド属性を設定する場合は、この要素に属性を明示的に設定する必要があります。</p> <p>個々のパラメータの詳細については、「<b>ステートメントのパラメータ</b>」の表を参照してください。</p> <p><b>注意:</b> データベースフィールドに編集マスクが定義されている場合は、セッションパラメータ EM が DDM で動的に参照されます。編集マスクは出力フィールドと入力フィールドに対してのみ指定できます。編集マスクが入力フィールドに対して定義された場合、フィールドのデータは編集マスクの仕様に従って入力する必要があります。</p>	

属性

次の属性を使用できます。



1. 表示属性：『パラメータリファレンス』でセッションパラメータ AD を参照してください。
2. 色属性：『パラメータリファレンス』でセッションパラメータ CD を参照してください。
3. アウトライン属性：『パラメータリファレンス』でセッションパラメータ BX を参照してください。
4. 出力モード属性：『パラメータリファレンス』でセッションパラメータ PM を参照してください。

## 例 - 構文 1

- 例 1 - INPUT ステートメント
- 例 2 - DEFINE WINDOW ステートメントのある INPUT ステートメント
- 例 3 - MARK POSITION オプションのある INPUT ステートメント

### 例 1 - INPUT ステートメント

```

** Example 'IPTEX1': INPUT
*****
DEFINE DATA LOCAL
1 #FNC (A1)
END-DEFINE
*
INPUT 10X 'SELECTION MENU FOR EMPLOYEES SYSTEM' /
      10X '-' (35) //
      10X 'ADD      (A)' /
      10X 'UPDATE  (U)' /
      10X 'DELETE  (D)' /
      10X 'STOP    (.)' //
      10X 'PLEASE ENTER FUNCTION: ' #FNC
*
DECIDE ON EVERY VALUE OF #FNC
  VALUE 'A' /* invoke the object containing the add function here
    WRITE 'Add function selected.'
  VALUE 'U' /* invoke the object containing the update function here
    WRITE 'Update function selected.'

```

## INPUT 構文 1- ダイナミック画面レイアウトの指定

```
VALUE 'D' /* invoke the object containing the delete function here
  WRITE 'Delete function selected.'
VALUE '.'
  STOP
NONE
  REINPUT 'Please enter a valid function.' MARK *#FNC
END-DECIDE
*
END
```

プログラム **IPTEX1** の出力：

```
SELECTION MENU FOR EMPLOYEES SYSTEM
-----
ADD      (A)
UPDATE  (U)
DELETE  (D)
STOP    (.)

PLEASE ENTER FUNCTION:
```

### 例 2 - DEFINE WINDOW ステートメントのある INPUT ステートメント

```
** Example 'INPEX1': INPUT (with DEFINE WINDOW statement)
*****
DEFINE DATA LOCAL
1 #STRING (A15)
END-DEFINE
*
DEFINE WINDOW WIND1
  SIZE 10 * 40
  BASE 5 / 10
  FRAMED ON POSITION TEXT
*
INPUT WINDOW='WIND1'
  'PLEASE ENTER HERE:' / #STRING
*
END
```

プログラム **INPEX1** の出力：

```
+-----Top+
! PLEASE ENTER HERE:      !
! #STRING                  !
!                          !
!                          !
!                          !
```

```

!
!
!
+-----Bottom+

```

### 例 3 - MARK POSITION オプションのある INPUT ステートメント

```

** Example 'INPEX2': INPUT (with POSITION)
*****
DEFINE DATA LOCAL
1 #START (A30)
END-DEFINE
*
ASSIGN #START = 'EXAM_'
*
INPUT (AD=M) MARK POSITION 5 IN *#START
      / 'PLEASE COMPLETE START VALUE FOR SEARCH'
      / 5X #START
END

```

プログラム **INPEX2** の出力：

```

PLEASE COMPLETE START VALUE FOR SEARCH
      #START EXAM[]

```



## 77 INPUT 構文 2 - 定義済みマップレイアウトの使用

---

■ パラメータリストのない INPUT USING MAP .....	518
■ プログラムで定義されている INPUT フィールド .....	519
■ INPUT 構文 2 - 説明 .....	519
■ 非スクリーンモードでの INPUT ステートメントの使用 .....	520
■ Natural スタックデータの処理 .....	522
■ バッチモードでの INPUT ステートメントの使用 .....	522

この形式の INPUT ステートメントは、Natural マップエディタを使用して作成されたマップレイアウトを使用して入力処理を実行するために使用されます。

マップレイアウトは次の 2 つの方法で使用できます。

- プログラムがパラメータリストを提供しない。
- プログラムがパラメータリスト (*operand1*) を提供する。

```
INPUT [WINDOW='window-name'] [WITH-TEXT-option]
[MARK-option]
[ALARM-option]
[USING] MAPmap-name [NO ERASE]
[
    operand1 ...
    NO PARAMETER
]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## パラメータリストのない INPUT USING MAP

---

パラメータリストのない INPUT USING MAP を使用する場合は、次の要件を満たす必要があります。

- *map-name* を英数字定数として指定する必要があります (最大 8 文字)。
- この方法で使用されるマップは、マップを参照するプログラムをコンパイルする前に作成する必要があります。
- 処理されるフィールドの名前は、コンパイル時にマップソース定義から動的に取得されます。使用されるフィールド名はプログラムとマップの両方で同一である必要があります。
- INPUT ステートメントで参照されるすべてのフィールドは、その時点でアクセス可能である必要があります。
- ストラクチャードモードでは、フィールドはあらかじめ定義されている必要があります (データベースフィールドは、処理ループまたはビューに正しく参照される必要があります)。
- レポートモードでは、マップにユーザー定義変数を新たに定義できます。
- マップレイアウトが変更された場合でも、マップを使用しているプログラムを再カタログする必要はありません。ただし、配列構造または名前、フィールドの形式/長さが変更された場合、またはマップでフィールドが追加/削除された場合は、マップを使用しているプログラムを再カタログする必要があります。
- マップソースはプログラムコンパイル時に使用可能である必要があります。そうでないと、INPUT USING MAP ステートメントをコンパイルできません。





**注意:** マップがまだ使用可能でない場合にプログラムをコンパイルするには、NO PARAMETER を指定します。これにより、マップがまだ使用可能でなくても INPUT USING MAP をコンパイルできます。

## プログラムで定義されている INPUT フィールド

プログラムで処理するフィールドの名前 (*operand1*) を指定することで、マップ内のフィールドの名前とは異なる名前をプログラムのフィールドの名前にすることができます。

プログラムのフィールドの順序はマップの順序と一致している必要があります。マップエディタは、マップで指定されたフィールドを、フィールド名の英字順でソートすることに注意してください。詳細については、Natural の『エディタ』ドキュメントのマップエディタの説明を参照してください。

プログラムエディタの行コマンド `.I(mapname)` を使用して、指定されたマップで定義されているフィールドから派生したパラメータリストとともに完全な INPUT USING MAP ステートメントを取得できます。

マップのレイアウトが変更された場合でも、マップを使用してプログラムを再カタログする必要はありません。ただし、フィールド名、フィールド形式/長さ、またはマップ内の配列構造が変更された場合、またはマップでフィールドが追加または削除された場合は、プログラムを再カタログする必要があります。

プログラムで指定されたフィールドの形式と長さが、マップで指定されたフィールドと一致することを確認するために、実行時にチェックが行われます。両方のレイアウトが一致しない場合は、エラーメッセージが生成されます。

## INPUT 構文 2 - 説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>map-name</i>	C S	A U	可	不可
<i>operand1</i>	S A	A U N P I F B D T L C	可	可

構文要素の説明：

<b>INPUT WINDOW='window-name'</b>	このオプションについては、INPUT ステートメントの「 <a href="#">構文 1</a> 」で説明しています。
<b>WITH TEXT/MARK/ALARM-options</b>	これらのオプションについては、INPUT ステートメントの「 <a href="#">構文 1</a> 」で説明しています。「 <a href="#">WITH TEXT オプション</a> 」、「 <a href="#">MARK オプション</a> 」、「 <a href="#">ALARM オプション</a> 」を参照してください。
<b>USING MAP map-name</b>	<p>USING MAP は、マップエディタを使用して以前に Natural システムファイルに格納されたマップ定義を呼び出します。</p> <p>map-name は、1~8 文字の英数字定数またはユーザー定義変数にすることができます。変数を使用する場合、事前にこれを定義しておく必要があります。マップ名にアンパサンド (&amp;) を含めることができます。この文字は実行時に Natural システム変数 *LANGUAGE の現在の値で置き換えられます。この機能で複数言語のマップを使用できます。</p> <p>INPUT ステートメントを実行すると、NO ERASE オプションが指定されていない限り、対応するマップで画面の現在の内容が置き換えられます。このオプションを指定した場合は、マップが画面の現在の内容をオーバーレイします。</p>
<b>NO ERASE</b>	このオプションについては、INPUT ステートメントの「 <a href="#">構文 1</a> 」で説明しています。
<b>operand1</b>	<p>フィールド指定：</p> <p>データベースフィールドまたはユーザー定義変数、あるいはその両方のリスト。フィールドは、参照先のマップ内のフィールドと、数、順序、形式、長さ、およびオカレンス数（配列の場合）が一致している必要があります。そうでないとエラーが発生します。</p> <p>データベースフィールドの内容が INPUT 処理の結果として変更される場合は、データエリアにある値のみが変更されます。適切なデータベースステートメント <a href="#">UPDATE/STORE</a> を使用して、データベースの内容を変更する必要があります。</p>

## 非スクリーンモードでの INPUT ステートメントの使用

セッションパラメータ IM または端末コマンド %F および %D を使用して、入力モードを変更できます。

## フォームモード

端末コマンド %F を実行すると、フォームモードが有効になります。

フォームモード（プロファイル/セッションパラメータ IM=F）では、Natural は位置パラメータに従ってフィールドごとに端末にマッピングアウトの出力テキストをすべて表示します。これにより、ユーザーはフィールドごとにデータを入力できます。すべてのデータが入力されると、ハードコピーの出力が、画面に表示されるのと同じ形式で生成されます。

フォームモードでは、%R と入力すると、エラーの発生時にオペレータがフォーム全体を再入力できます。入力は、INPUT ステートメントの最初の実行時に処理されます。

## キーワード/デリミタモード

端末コマンド %D を実行すると、キーワード/デリミタモードが有効になります。

キーワード/デリミタモード（プロファイル/セッションパラメータ IM=D）では、キーワードまたは位置入力値を使用してデータを入力できます。

キーワード入力を使用すると、端末オペレータは、プロンプトテキストを使用して個々のフィールドのデータを入力できます。フォームモードでは、プロンプトテキストは、フィールドを識別するキーワードとして値の前に表示されます。キーワードの後には、INPUT 割り当て文字（IA パラメータ）、その直後にデータが続く必要があります。割り当て文字に続くスペースは、区切り文字（ID パラメータ）までデータとして取得されます。区切り文字は、最後のデータ要素の後には不要です。異なるフィールドのキーワードデータは、区切り文字で分割して任意の順序で入力できます。INPUT ステートメントで定義されていないキーワードをオペレータが入力した場合は、エラーメッセージが返されます。入力フィールドにデータを入力する必要はありません。データが入力されていないフィールドは、英数字フィールドでは空白に、数値および 16 進数フィールドではゼロに設定されます。

位置値入力を使用する場合、端末オペレータは、現在定義されている INPUT 区切り文字（ID パラメータ）で区切られたすべての入力フィールドのデータのみ入力します。入力するフィールドの順序は、INPUT ステートメントのフィールドの順序に対応します。

ユーザーは、デリミタ文字で区切った位置入力に値の数を入力し、値の前にキーワードを指定して選択したフィールドのキーワードモードに切り替えることにより、位置入力からキーワード入りに切り替えることができます。

キーワードがフィールドの位置指定に使用された場合、キーワードの値に続くキーワード以外の入力は、INPUT ステートメントで以前に選択されたフィールドに続くフィールドに割り当てられる位置入力として処理されます。



**注意:** キーワードおよび対応する入力フィールドは、同じ論理行にある必要があります。合計の長さが行サイズを超える場合は、行サイズ（LS パラメータ）を適宜調整して、キーワードとフィールドが 1 行に収まるようにします。

キーワード／デリミタモードで入力されたデータは、スクリーンモードの場合と同じように検証されます。フィールドに対して定義されている文字数を超過して入力しようとした場合は、エラーメッセージが返されます。

バッファ（3270タイプ）端末またはワークステーションで、INPUT ステートメントをキーワード／デリミタモードで処理する場合は、1つの INPUT ステートメントに割り当てられるすべてのデータを1つの画面に入力する必要があります。Enter は、INPUT ステートメントに対するすべてのデータが入力された場合にのみ使用します。

## Natural スタックデータの処理

---

Natural スタックに **FETCH**、**RUN**、または **STACK** ステートメントを介して配置されたデータ要素は、実行時に検出された次の INPUT ステートメントで処理されます。

INPUT ステートメントは、前述したようにキーワード／デリミタモードのデータを処理します。

すべての入力フィールドを充填するためのデータ要素が使用可能でない場合は、フィールド形式に応じて空白／ゼロが充填されます。存在する入力フィールドよりも多くのデータ要素が指定されている場合、残りのデータは無視されます。

フィールドにスタックからのデータが入力されている場合、フィールド属性はデータに適用されません。

Natural システム変数 \*DATA を参照して、Natural スタックで現在使用可能なデータ要素の数を判断できます。

## バッチモードでの INPUT ステートメントの使用

---

以下では次のトピックについて説明します。

- **バッチフォームモード**
- **バッチキーワード／デリミタモード**

## ■ バッチモードでの端末コマンドの使用

### バッチフォームモード

バッチフォームモードでは、INPUT マップが表示されます。データレコードは、1つ以上の AD=A フィールドや AD=M フィールドを含む行ごとに読み取られ、レコードに含まれるデータが1つ以上の適切なフィールドに割り当てられます。

入力データフィールドは、連続していると想定されます。デリミタ文字が使用されていない限り、入力データはフィールドの内部定義に従って正確な長さで入力する必要があります。数値フィールドの場合は、記号 (SG=ON の場合) および小数点に空白を適宜許可する必要があります。

オプションとして、個々のフィールドの値を分割するデリミタ文字を使用してデータを入力することもできます。この場合、内部定義に従って正確な位置番号で入力する必要はなく、位置 1 から開始して左から右に処理されます。データ入力のルールは、「[データ入力](#)」で説明したルールと同じです。また、割り当て文字を使用して、\*OUTIN フィールドの内容がリセットされないことを指定できます。

### バッチキーワード／デリミタモード

キーワード／デリミタモードがバッチモードで使用されている場合は、TP モードのキーワード／デリミタモードと同じように機能しますが、次の例外があります。

- 入力マップ全体を端末コマンド %Q の制御下で出力できます。
- \*OUTIN フィールドは、明示的に変更されない限り元の値を保持します。

### バッチモードでの端末コマンドの使用

次の Natural 端末コマンドは、INPUT ステートメントをメインフレームコンピュータでバッチモードで使用している場合に使用できます。

コマンド	説明
%*	レコード抑制。レコードの位置 1 および 2 に入力された場合、%* を使用すると次の入力レコードの出力が抑制されます。
	DATA RECORD %* SUPPRESSED DATA RECORD

コマンド	説明
%	<p>レコード継続。%がレコードの最後の空白以外の文字として入力された場合は、次の入力レコードが継続レコードとして扱われます。</p> <pre>DATA, RECORD, WITH, CONTINUATION, % CONTINUATION RECORD</pre> <pre>INPUT V1 V2 V3 V4 V5 V6 DISPLAY V1 V2 V3 V4 V5 V6</pre> <p>次の出力を生成します。</p> <pre>DATA RECORD WITH CONTINUATION CONTINUATION RECORD</pre>
%/	<p>エンドオブファイル。レコードの最初の2つの位置に入力された場合（最後に空白以外の文字はなし）、%/ はエンドオブファイル条件を発生させます。</p>
%%	<p>入力データストリームに再開位置を設定します。</p>
%.	<p>現在の INPUT ステートメントの入力値の読み取りが終了します。</p>
%Knn	<p>PF キーをシミュレートします。</p>
%KPn	<p>PA キーをシミュレートします。</p>
%Q	<p>このコマンドでは、入力データの読み取りに使用されるマップの出力が抑制されます。</p>

詳細については、『端末コマンド』ドキュメントを参照してください。

バッチモードでのデータ入力に INPUT ステートメントを使用している場合は、追加の JCL ステートメントが必要です。Natural をバッチモードで実行する前に、これらのステートメントが提供されていることを Natural 管理者に確認する必要があります。

# 78 INTERFACE

---

▪ 機能 .....	526
▪ 構文説明 .....	527

```
INTERFACE interface-name
  [property-definition]
  [method-definition]
END-INTERFACE
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CREATE OBJECT](#) | [DEFINE CLASS](#) | [INTERFACE](#) | [METHOD](#) | [PROPERTY](#) | [SEND METHOD](#)

関連機能グループ：「[コンポーネントベースプログラミング](#)」

## 機能

---

インターフェイスは、互いに意味のあるメソッドおよびプロパティの集まりであり、クラスの特  
定機能を提供します。

クラスに対して1つ以上のインターフェイスを定義することができます。複数インターフェイスを定義すると、それらが行うことに従ってメソッドを構造化／グループ化することが可能です。例えば、すべてのメソッドを挿入し、1つのインターフェイスの持続性（ロード、ストア、更新）を扱い、他のメソッドを他のインターフェイスに挿入します。

INTERFACE ステートメントは、インターフェイスを定義するために使用します。Natural クラスモジュールでのみ使用し、次のように定義できます。

- [DEFINE CLASS](#) ステートメント内で定義します。この形式は、インターフェイスが1つのクラスに実装されるときにのみ使用します。
- または、[DEFINE CLASS](#) ステートメントの [INTERFACE USING](#) 節に組み込まれるコピーコードで定義します。この形式は、インターフェイスが複数のクラスに実装されるときに使用します。

インターフェイスに関連するプロパティとメソッドは、プロパティとメソッド定義で定義します。



## 構文説明

<i>interface-name</i>	<p>インターフェイスに割り当てる名前です。インターフェイス名は最大 32 文字以内で、ユーザー定義変数の命名規則に従っている必要があります（『<i>Natural</i> の使用』ドキュメントの「ユーザー定義変数の命名規則」を参照）。クラスごとに一意で、クラス名とも異なっている必要があります。</p> <p>異なるプログラミング言語で書かれたクライアントによってインターフェイスを使用する予定であれば、インターフェイス名はこれらの言語に適用する命名規則と矛盾しないように選択する必要があります。</p>
<i>property-definition</i>	プロパティ定義は、インターフェイスのプロパティを定義するために使用します。下記の「 <a href="#">プロパティ定義</a> 」を参照してください。
<i>method-definition</i>	メソッド定義は、インターフェイスのメソッドを定義するために使用します。下記の「 <a href="#">メソッド定義</a> 」を参照してください。
<b>END-INTERFACE</b>	INTERFACE ステートメントを終了するには、 <i>Natural</i> の予約語 END-INTERFACE を使用する必要があります。

### プロパティ定義

プロパティ定義は、インターフェイスのプロパティを定義するために使用します。

```
PROPERTY property-name
  [(format-length/array-definition)]
  [READONLY]
  [ISoperand]
END-PROPERTY
```

プロパティは、クライアントによってアクセス可能なオブジェクトの属性です。例えば、従業員を表すオブジェクトは Name プロパティと Department プロパティを持ちます。Name または Department プロパティをアクセスすることによって従業員の名前や部署を検索したり変更したりするほうが、値を返す1つのメソッドおよび値を変更する別のメソッドを呼び出すより、クライアントにとっては非常に簡単です。

各プロパティは、その値を保存するためにクラスのオブジェクトデータエリアに変数を必要とします。これはオブジェクトデータ変数として参照されます。プロパティ定義を使用してこの変数をクライアントにアクセス可能にします。プロパティ定義は、プロパティの名前とフォーマットを定義し、オブジェクトデータ変数に接続します。最も単純なケースでは、プロパティは、オブジェクトデータ変数自体の名前とフォーマットを使用します。一定の制限内の名前とフォーマットを変更することも可能です。

<p><i>property-name</i></p>	<p>プロパティに割り当てる名前です。プロパティ名は最大 32 文字以内で、Natural ユーザー定義変数の命名規則に従う必要があります。『Natural の使用』ドキュメントの「ユーザー定義変数の命名規則」を参照してください。</p> <p>異なるプログラミング言語で書かれたクライアントによってプロパティを使用する予定であれば、プロパティ名はこれらの言語に適用する命名規則と矛盾しないように選択する必要があります。</p>
<p><i>format-length/array-definition</i></p>	<p>クライアントに認識されるように、プロパティのフォーマットを定義します。</p> <p><i>format-length/array-definition</i> を省略すると、フォーマット長さと配列定義は、IS 節で割り当てられたオブジェクトデータ変数から取得されます。</p> <p><i>format-length/array-definition</i> を指定する場合、IS 節の <i>operand</i> に指定されたオブジェクトデータ変数のフォーマットとの転送の互換性が必要です。READONLY プロパティの場合、データ転送の互換性は 1 つの方向（ソースオペランドとしてのオブジェクトデータ変数と宛先オペランドとしてのプロパティ）にのみ必要です。配列定義を指定するときは、次元、次元ごとのオカレンス、下限、および上限を、対応するオブジェクトデータ変数の配列定義と等しくする必要があります。これは、次元ごとに 1 つのアスタリスクを指定することによって表現されます。</p>
<p>READONLY</p>	<p>このキーワードが指定されると、プロパティの値は読まれるだけで、設定されません。IS 節の <i>operand</i> に指定されたオブジェクトデータ変数のフォーマットは、<i>format-length/array-definition</i> に指定されたフォーマットへのデータ転送の互換性が必要です。逆方向へのデータ転送の互換性は必要ありません。</p> <p>キーワード READONLY が省略されると、プロパティの値は読み込みおよび設定が可能です。</p>
<p>IS operand</p>	<p>IS 節の <i>operand</i> により、オブジェクトデータ変数をプロパティ値の保存場所として割り当てます。割り当てられたオブジェクトデータ変数はグループでなくてもかまいません。変数は正常なオペランド構文で参照されます。これは、オブジェクトデータ変数が配列であれば、添字指定で参照する必要があることを意味しています。完全な添字範囲指定とアスタリスク表記だけが許されます。</p>

	<p>INTERFACE ステートメントがコピーコードメンバから組み込まれ、複数のクラスで再利用されるならば、IS 節は使用されません。INTERFACE ステートメントを再利用したい場合は、INTERFACE ステートメント外で <b>PROPERTY</b> ステートメントにオブジェクトデータ変数を割り当てる必要があります。</p> <p>IS 節を省略すると、プロパティはプロパティと同じ名前のオブジェクトデータ変数と接続されます。この名前を持つ変数が定義されていないか、または、それがグループであれば、構文エラーが生じます。</p>
<b>END-PROPERTY</b>	<p><b>Natural</b> 予約語 END-PROPERTY を使用して、インターフェイス PROPERTY 定義を終了させる必要があります。</p>

## 例

オブジェクトデータエリアに以下のデータ定義を含ませてください。

```
1 Salary(p7.2)
1 SalaryHistory(p7.2/1:10)
```

その後、以下のプロパティ定義が可能になります。

```
property Salary
  end-property
property Pay is Salary
  end-property
property Pay(P7.2) is Salary
  end-property
property Pay(N7.2) is Salary
  end-property
property SalaryHistory
  end-property
property OldPay is SalaryHistory(*)
  end-property
property OldPay is SalaryHistory(1:10)
  end-property
property OldPay(P7.2/*) is SalaryHistory(1:10)
  end-property
property OldPay(N7.2/*) is SalaryHistory(*)
  end-property
```

以下のプロパティ定義は許されません。

```

/* Not data transfer-compatible. */
property Pay(L) is Salary
end-property
/* Not data transfer-compatible. */
property OldPay(L/*) is SalaryHistory(*)
end-property
/* Not data transfer-compatible. */
property OldPay(L/1:10) is SalaryHistory(1:10)
end-property
/* Assigns an array to a scalar. */
property OldPay(P7.2) is SalaryHistory(1:10)
end-property
/* Takes only a sub-array. */
property OldPay(P7.2/3:5) is SalaryHistory(*)
end-property
/* Index specification omitted in ODA variable SalaryHistory. */
property OldPay is SalaryHistory
end-property
/* Only asterisk notation allowed in property format specification. */
property OldPay(P7.2/1:10) is SalaryHistory(*)
end-property

```

## メソッド定義

メソッド定義は、インターフェイスのメソッドを定義するために使用します。

```

METHOD method-name
  [IS subprogram-name]
  [
    PARAMETER { USING parameter-data-area } ] ...
  [data-definition ... ]
END-METHOD

```

インターフェイスを各種クラスで再利用可能にするためには、コピーコードからインターフェイス定義を組み込み、インターフェイス定義の後に METHOD ステートメントでサブプログラムを定義してください。その後、各種クラスでそれぞれにメソッドを実装できます。

<i>method-name</i>	<p>メソッドに割り当てる名前です。メソッド名は最大 32 文字で、<i>Natural</i> の命名規則に従う必要があります（『<i>Natural</i> の使用』ドキュメントの「ユーザー定義変数の命名規則」を参照）。インターフェイスごとに一意にする必要があります。</p> <p>異なるプログラミング言語で書かれたクライアントによってメソッドを使用する予定であれば、メソッド名はこれらの言語に適用する命名規則と矛盾しないように選択する必要があります。</p>
--------------------	---

<b>IS</b> <i>subprogram-name</i>	<p>これは、メソッドを実装するサブプログラムの名前です。サブプログラムの名前は最大8文字です。デフォルトはメソッド名です（IS節が指定されない場合）。</p>
<b>PARAMETER</b>	<p>メソッドのパラメータを指定します。DEFINE DATA ステートメントの PARAMETER 節と同じ構文です。</p> <p>パラメータは、サブプログラムの実装において後で使用されるパラメータと一致している必要があります。これは、パラメータデータエリアを使用することによって保証されます。</p> <p>パラメータデータエリアの BY VALUE とマークされたパラメータは、メソッドの入力パラメータです。</p> <p>BY VALUE が指定されていないパラメータは、「by reference」（参照）で渡される入力/出力パラメータです。これがデフォルトです。</p> <p>BY VALUE RESULT とマークされている最初のパラメータが、メソッドの戻り値として返されます。複数のパラメータがこのようにマークされている場合は、その他のパラメータは入力/出力パラメータとして扱われます。</p> <p>メソッドが呼び出される場合は、OPTIONAL パラメータを指定する必要はありません。SEND METHOD ステートメントで <i>nX</i> 表記を使用することにより、これらを未指定にしておくことができます。</p>
<b>END-METHOD</b>	<p>Natural 予約語 END-METHOD を使用して、インターフェイスの METHOD 定義を終了させる必要があります。</p>



# 79

## LIMIT

---

■ 機能 .....	534
■ 構文説明 .....	535
■ 例 .....	535

LIMIT *n*

関連ステートメント：ACCEPT/REJECT | AT BREAK | AT START OF DATA | AT END OF DATA | BACKOUT TRANSACTION | BEFORE BREAK PROCESSING | DELETE | END TRANSACTION | FIND | GET | GET SAME | GET TRANSACTION | HISTOGRAM | PASSW | PERFORM BREAK PROCESSING | READ | RETRY | STORE | UPDATE

関連機能グループ：「データベースへのアクセスと更新」

## 機能

LIMIT ステートメントは、FIND、READ または HISTOGRAM ステートメントで開始された処理ループの実行回数を制限するために使用します。

制限値は、他の LIMIT ステートメントで上書きされるまで、プログラム内の後続のすべての処理ループに有効となります。

LIMIT ステートメントは、制限値が明示されている個々のステートメント（例：FIND(*n*) ...）には適用されません。

制限値に達すると処理が終了し、メッセージが表示されます。また、処理ループが制限値を超えた場合の反応を決定するセッションパラメータ LE も参照してください。

LIMIT ステートメントの指定がなければ、Natural インストール時に Natural プロファイルパラメータ LT と一緒に定義されたデフォルトのグローバル制限が使用されます。

### レコードのカウント

処理ループが制限に達したかどうかを確定するために、ループ内で読み込んだ各レコードを制限に対してカウントします。処理ループの制限には以下が適用されます。

- FIND や READ ステートメントの WHERE 節で除かれたレコードは、制限数の対象にはなりません。
- ACCEPT/REJECT ステートメントで排除されたレコードは、制限数にカウントされます。



## 構文説明

<b>LIMIT</b> <i>n</i>	<p>制限値の指定</p> <p>制限値 <i>n</i> は 0~4294967295（リーディングゼロはオプション）の数値定数で指定します。</p> <p>制限値がゼロであれば、処理ループに入りません。</p>
-----------------------	---

## 例

- 例 1 - LIMIT ステートメント
- 例 2 - LIMIT ステートメント（2つのデータベースループに対して有効）

### 例 1 - LIMIT ステートメント

```

** Example 'LMTEX1': LIMIT
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 CITY
END-DEFINE
*
LIMIT 4
*
READ EMPLOY-VIEW BY NAME STARTING FROM 'BAKER'
  DISPLAY NOTITLE
    NAME PERSONNEL-ID CITY *COUNTER
END-READ
*
END

```

プログラム **LMTEX1** の出力：

NAME	PERSONNEL ID	CITY	CNT
BAKER	20016700	OAK BROOK	1
BAKER	30008042	DERBY	2
BALBIN	60000110	BARCELONA	3
BALL	30021845	DERBY	4

**例 2 - LIMIT ステートメント (2つのデータベースループに対して有効)**

```
** Example 'LMTEX2': LIMIT (valid for two database loops)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
END-DEFINE
*
LIMIT 3
*
FIND EMPLOY-VIEW WITH NAME > 'A'
  READ EMPLOY-VIEW BY NAME STARTING FROM 'BAKER'
    DISPLAY NOTITLE 'CNT(0100)' *COUNTER(0100)
                      'CNT(0110)' *COUNTER(0110)
  END-READ
END-FIND
*
END
```

プログラム **LMTEX2** の出力：

CNT(0100)	CNT(0110)
1	1
1	2
1	3
2	1
2	2
2	3
3	1
3	2
3	3

# 80 LOOP

---

▪ 機能 .....	538
▪ 制限事項 .....	538
▪ 構文説明 .....	538
▪ 例 .....	539

[CLOSE] LOOP [(r)]
--------------------

## 機能

LOOP ステートメントは、処理ループを終了するために使用します。ループの現在のパスの処理が終了し、制御が処理ループの先頭に戻ります。

LOOP ステートメントの発行対象の処理ループが終了した場合（つまり、すべてのレコードが処理されるか、反復が実行された場合）は、実行が LOOP ステートメントの後のステートメントから続行されます。

### データベース変数参照

処理ループの終了に加えて、LOOP ステートメントは、ループ内に含まれる **FIND**、**FIND FIRST**、**FIND UNIQUE**、**READ** および **GET** ステートメントへのすべてのフィールド参照を除外します。


ビュー内のフィールドは、ビュー名を修飾子として使用して、処理ループの外部で参照できません。

## 制限事項

- 当ステートメントはレポーティングモードでだけ有効です。
- LOOP ステートメントは、**IF** や **AT BREAK** などの条件付きステートメントに基づいて指定することはできません。

## 構文説明

<b>LOOP (r)</b>	ステートメント参照表記：  LOOP ステートメントは、ステートメントラベルまたは参照番号（表記 (r)）で指定できます。その場合、参照されているステートメントによって開始されたループまで（そのループを含む）のすべての内部ループが終了します。ステートメント参照が指定されていない場合は、最も内側のアクティブな処理ループが終了します。
-----------------	--

 **注意:** レポーティングモードでは、現在アクティブになっているすべての処理ループ（LOOP ステートメントで明示的に終了していないもの）が **END** ステートメントにより自動的に終了します。

---

## 例

---

### 例 1

```
FIND ...  
  READ ...  
    READ ...  
LOOP (0010)  /* closes all loops
```

### 例 2

```
FIND ...  
  READ ...  
    READ ...  
      LOOP      /* closes loop initiated on line 0030  
    LOOP      /* closes loop initiated on line 0020  
  LOOP      /* closes loop initiated on line 0010
```



# 81 METHOD

---

■ 機能 .....	542
■ 構文説明 .....	542
■ 例 .....	543

```

METHOD method-name
  OF [INTERFACE] interface-name
  IS subprogram-name
END-METHOD

```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CREATE OBJECT](#) | [DEFINE CLASS](#) | [INTERFACE](#) | [PROPERTY](#) | [SEND METHOD](#)

関連機能グループ：「[コンポーネントベースプログラミング](#)」

## 機能

METHOD ステートメントは、実装としてサブプログラムをメソッドにインターフェイス定義外で割り当てます。問題のインターフェイス定義がコピーコードから取得され、クラス固有の方法で実装される場合は、これが使用されます。

METHOD ステートメントは、[DEFINE CLASS](#) ステートメント内でインターフェイス定義後のみ使用できます。指定されたインターフェイスとメソッドの名前は、[DEFINE CLASS](#) ステートメントの [INTERFACE](#) 節で定義されている必要があります。

## 構文説明

<i>method-name</i>	これは、 <a href="#">メソッド</a> に割り当てられた名前です。
OF <i>interface-name</i>	これは、 <a href="#">インターフェイス</a> に割り当てられた名前です。
IS <i>subprogram-name</i>	これは、メソッドを実装するサブプログラムの名前です。サブプログラムの名前は最大 8 文字です。デフォルトは <i>method-name</i> です（IS 節が指定されない場合）。
END-METHOD	METHOD ステートメントを終了するには、Natural の予約語 END-METHOD を使用する必要があります。




## 例

以下の例は、同じインターフェイスを2つのクラスに別々に実装する方法と、これを実現するために **PROPERTY** ステートメントと **METHOD** ステートメントを使用する方法を示しています。

インターフェイス Measure をコピーコード iface-c に定義します。クラス Elephant と Mouse はインターフェイス Measure を実装します。したがって、それらは両方ともコピーコード iface-c を組み込みます。しかし、クラスは、それらの個々のオブジェクトデータエリアから異なる変数を用いてプロパティ Height を実装し、異なるサブプログラムでメソッド Init を実装します。それらは **PROPERTY** ステートメントを使用して選択したデータエリア変数をプロパティに割り当てて、**METHOD** ステートメントで選択したサブプログラムをメソッドに割り当てます。

プログラム prog は両方のクラスのオブジェクトを作成し、個々のクラス実装への初期化の指定をそのままにして、同じメソッド Init を使用してそれらを初期化できるようになります。



次は上記の例で使用された Natural モジュールの完全な内容を示しています。

コピーコード : **iface-c**

```
interface Measure
*
property Height(p5.2)
end-property
*
property Weight(i4)
end-property
*
method Init
end-method
*
end-interface
```

クラス : **class1**

```
define class elephant
object using class1-o
interface using iface-c
*
property Height of interface Measure is height
end-property
*
property Weight of interface Measure is weight
end-property
*
method Init of interface Measure is init1-n
end-method
*
end-class
end
```

オブジェクトデータエリア : **class1-o**

```
*   *** Top of Data Area ***
  1 HEIGHT                P 5.2
  1 WEIGHT                 I 2
*   *** End of Data Area ***
```

メソッドサブプログラム : **init1-n**

```
define data
object using class1-o
end-define
*
height := 17.3
weight := 120
*
end
```

## クラス：**class2**

```
define class mouse
object using class2-o
interface using iface-c
*
property Height of interface Measure is size
end-property
*
property Weight of interface Measure is weight
end-property
*
method Init of interface Measure is init2-n
end-method
*
end-class
end
```

## オブジェクトデータエリア：**class2-o**

```
*   *** Top of Data Area ***
1  SIZE                P 3.2
1  WEIGHT              I 1
*   *** End of Data Area ***
```

## メソッドサブプログラム：**init2-n**

```
define data
object using class2-o
end-define
*
size := 1.24
weight := 2
*
end
```

## プログラム：**prog**

```
define data local
1 #o handle of object
1 #height(p5.2)
1 #weight(i4)
end-define
*
create object #o of class 'Elephant'
send "Init" to #o
#height := #o.Height
#weight := #o.Weight
write #height #weight
*
```

## METHOD

---

```
create object #o of class 'Mouse'  
send "Init" to #o  
#height := #o.Height  
#weight := #o.Weight  
write #height #weight  
*  
end
```

# 82 MOVE

---

▪ 機能 .....	548
▪ 構文説明 .....	549
▪ 例 .....	560

関連ステートメント：[ADD](#) | [COMPRESS](#) | [COMPUTE](#) | [DIVIDE](#) | [EXAMINE](#) | [MOVE ALL](#) | [MULTIPLY](#) | [RESET](#) | [SEPARATE](#) | [SUBTRACT](#)

関連機能グループ：「[算術演算とデータ移動操作](#)」

## 機能

MOVE ステートメントは、オペランドの値を1つ以上のオペランド（フィールドまたは配列）に移動するために使用します。

複数のターゲットオペランドを持つ MOVE ステートメントは、対応する個々の MOVE ステートメントと同一です。

```
MOVE #SOURCE TO #TARGET1 #TARGET2
```

これは、以下と同じです。

```
MOVE #SOURCE TO #TARGET1
MOVE #SOURCE TO #TARGET2
```

例：

```
DEFINE DATA LOCAL
1 #ARRAY(I4/1:3) INIT <3,0,9>
1 #INDEX(I4)
1 #RESULT(I4)
END-DEFINE
*
#INDEX := 1
MOVE #ARRAY(#INDEX) TO #INDEX      /* #INDEX is 3
                        #RESULT     /* #RESULT is 9
*
#INDEX := 2
MOVE #ARRAY(#INDEX) TO #INDEX      /* #INDEX is 0
                        #ARRAY(3)  /* returns run time error NAT1316
```

*operand2* が動的変数変数のとき、その長さは MOVE 操作によって変更できます。動的変数の現在の長さは、システム変数 \*LENGTH を使用して確認できます。動的変数については、『プログラミングガイド』の「[ラージ変数／フィールドと動的変数／フィールド](#)」を参照してください。

*operand2* のフォーマットが C の場合、*operand1* は (*parameter*) として指定することもできます。有効なパラメータは以下のとおりです。

MOVE ステートメントで指定可能なパラメータ		指定 (S=ステートメントレベル、E=要素レベル)
AD	属性定義	SE
CD	カラー定義	S


データ転送の互換性および規則についての詳細は、『プログラミングガイド』の「データ転送」を参照してください。

## その他の考慮事項

受け取りフィールドにデータベースフィールドが指定されても、MOVE 処理はプログラム内のフィールドの内部値を更新するだけで、データベースのフィールドの値は変更しません。

AT BREAK、AT END OF DATA または AT END OF PAGE ステートメントとともに MOVE ステートメントが指定された場合にのみ Natural システム関数が使用できます。

『プログラミングガイド』の「演算割り当てのルール」も参照してください。

 **注意:** *operand1* が時間変数 (フォーマット T) のとき、変数内容の時間コンポーネントだけが転送され、日付コンポーネントは転送されません (構文4と構文5で説明されている MOVE EDITED の場合を除く)。

## 構文説明

このステートメントには異なる構造が可能です。

- 構文 1 - MOVE ROUNDED
- 構文 2 - MOVE SUBSTRING
- 構文 3 - MOVE BY NAME / POSITION
- 構文 4 - MOVE EDITED (*operand2* で指定された編集マスク)
- 構文 5 - MOVE EDITED (*operand1* で指定された編集マスク)
- 構文 6 - MOVE LEFT/RIGHT JUSTIFIED
- 構文 7 - MOVE NORMALIZED
- 構文 8 - MOVE ENCODED

次の構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 構文 1 - MOVE ROUNDED

`MOVE [ROUNDED] operand1 [(parameter)] TO operand2 ...`

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S A N A U N P I F B D T L C G O		可	不可
<i>operand2</i>	S A M A U N P I F B D T L C G O		可	可

構文要素の説明：

<b>MOVE ROUNDED</b>	<p>当オプションは、<i>operand2</i> を四捨五入します。</p> <p>ROUNDED が指定できるのは <i>operand2</i> が数字タイプのときだけです。</p> <p><i>operand2</i> のフォーマットが N または P で、<i>operand2</i> が複数回指定されている場合は、ROUNDED は、小数点後の 7 桁を持つターゲットオペランドについては無視されます。</p> <p>「例 1 - MOVE ステートメントのさまざまな使用例」も参照してください。</p>	
<i>(parameter)</i>	<p><i>parameter</i> として、オプション PM=I またはセッションパラメータ DF を指定できます。</p>	
	PM=I	<p>右から左に記述する言語をサポートするために、PM=I を指定して、<i>operand1</i> の値を逆（右から左）方向で <i>operand2</i> に渡すことができます。</p> <p>例えば、次のステートメントの結果、#B の内容は ZYX となります。</p> <pre>MOVE 'XYZ' TO #A MOVE #A (PM=I) TO #B</pre> <p>PM=I は、<i>operand2</i> が英数字フォーマットのときにのみ指定できます。</p> <p><i>operand1</i> のトレーリングブランクは削除され（ブランクとバイナリゼロは削除され）、その後の値が逆向きに並べ替えられ、<i>operand2</i> に移されます。<i>operand1</i> が英数字フォーマットではない場合には、英数字フォーマットに変換されてから、逆向きに並べられます。</p> <p>PM=I と <a href="#">MOVE LEFT/RIGHT JUSTIFIED</a> も参照してください。</p>
	DF	<p><i>operand1</i> が日付変数で、<i>operand2</i> が英数字フィールドのとき、この日付変数に対するパラメータとして、セッションパラメータ DF を指定できます。セッションパラメータ DF については、『パラメータリファレンス』を参照してください。</p>



## 構文 2 - MOVE SUBSTRING

MOVE	{	<i>operand1</i> <u>SUBSTRING</u> ( <i>operand1</i> , <i>operand3</i> , <i>operand4</i> )	}	[( <i>parameter</i> )]	T0	{	<i>operand2</i> <u>SUBSTRING</u> ( <i>operand2</i> , <i>operand5</i> , <i>operand6</i> )	}	...
------	---	--	---	------------------------	----	---	--	---	-----

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S A	A U	可	不可
<i>operand2</i>	S A	A U	可	不可
<i>operand3</i>	C S	N P I B*	可	不可
<i>operand4</i>	C S	N P I B*	可	不可
<i>operand5</i>	C S	N P I B*	可	不可
<i>operand6</i>	C S	N P I B*	可	不可

\* 説明を参照してください。

構文要素の説明：

<b>MOVE SUBSTRING</b>	<p>SUBSTRING オプションの指定がないと、フィールドの内容全体が転送されます。</p> <p>SUBSTRING オプションで、英数字、Unicode またはバイナリフィールドの一部分だけを転送できます。SUBSTRING 節のフィールド名 (<i>operand1</i>) の後に、まず開始位置 (<i>operand3</i>)、次に移動するフィールド部分の長さ (<i>operand4</i>) を指定します。</p> <p><i>operand1</i> の基礎となるフィールドフォーマットが以下に該当する場合：</p> <ul style="list-style-type: none"> <li>■ 英数字 (A) またはバイナリ (B) の場合は、<i>operand3</i> または <i>operand4</i> で提供される値はバイト数と見なされます。</li> <li>■ Unicode (U) の場合は、<i>operand3</i> または <i>operand4</i> で提供される値は Unicode コード単位数、つまりダブルバイトと見なされます。</li> </ul> <p>例えば、フィールド #A の値の 5 桁目から 12 桁目をフィールド #B に転送するには、次のように指定します。</p> <pre>MOVE SUBSTRING(#A,5,8) TO #B</pre> <p><i>operand1</i> がダイナミック変数のとき、転送するフィールド部分は現在の長さ以下である必要があります。そうしないと、ランタイムエラーになります。</p> <p>また、受け取りフィールドの一部分に英数字、Unicode またはバイナリを転送することもできます。SUBSTRING 節のフィールド名 (<i>operand2</i>) の後に、まず開始位置 (<i>operand5</i>)、次に値を移動するフィールド部分の長さ (<i>operand6</i>) を指定します。</p>
---------------------------	--

*operand2* の基礎となるフィールドフォーマットが以下に該当する場合：

- 英数字 (A) またはバイナリ (B) の場合は、*operand5* または *operand6* で提供される値はバイト数と見なされます。
- Unicode (U) の場合は、*operand3* または *operand4* で提供される値は Unicode コード単位数、つまりダブルバイトと見なされます。

例えば、フィールド #A の値をフィールド #B の 3 桁目から 6 桁目に転送するには、次のように指定します。

```
MOVE #A TO SUBSTRING(#B,3,4)
```

*operand2* がダイナミック変数のとき、その開始位置 (*operand5*) は、現在の変数の長さに 1 を加えた長さ以下である必要があります。開始位置がその長さを超過すると、ランタイムエラーになります。これは、*operand2* の内容に確かでないギャップが生じるからです。

*operand3/5* または *operand4/6* がバイナリ変数の場合は、4 以下の長さでのみ使用できません。

*operand3/5* を省略すると、開始位置は 1 と見なされます。*operand4/6* を省略すると、長さはフィールドの開始位置から終わりまでの範囲とみなされます。

*operand2* がダイナミック変数であり、その指定開始位置 (*operand5*) が現在の変数の長さに 1 を加えた長さ (MOVE 処理が変数の長さを拡張するために使用されることを意味する) の場合、*operand6* は変数の新しい長さを決定するために指定する必要があります。

**注意:** SUBSTRING オプション指定の MOVE はバイトごとの転送です (『プログラミングガイド』の「算術演算の規則」に記述されている規則は適用されません)。

### 構文 3 - MOVE BY NAME / POSITION

```
MOVE BY { [NAME]
           POSITION } operand1 TO operand2
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	G		可	不可
<i>operand2</i>	G		可	不可

構文要素の説明：

<p><b>MOVE BY NAME</b> <i>operand1 TO</i> <i>operand2</i></p>	<p>当オプションは、あるデータ構造内の個々のフィールドを、その構造内の位置に関係なく別のデータ構造へと転送します。</p> <p>ただし、フィールド名が両方の構造に存在する場合にのみ転送されます。これは、再定義で作成されたフィールドと同様に再定義されたフィールドも含まれます。個々のフィールドは、どのフォーマットでもかまいません。オペランドにビューを指定することもできます。</p> <p><b>注意:</b> 個々の MOVE の順番は、<i>operand1</i> のフィールド順で決まります。</p> <p>「<a href="#">例 2 - MOVE BY NAME ステートメント</a>」も参照してください。</p> <p>配列を伴う <b>MOVE BY NAME</b></p> <p>データ構造が配列を持つ場合、MOVE 時その配列に添字 (*) が割り当てられます。その配列が割り当て処理の規則に合わない場合は、エラーを返します（『プログラミングガイド』の「<a href="#">配列処理</a>」を参照）。</p> <p>「<a href="#">例 3 - 配列を伴う MOVE BY NAME</a>」も参照してください。</p>
<p><b>MOVE BY POSITION</b> <i>operand1 TO</i> <i>operand2</i></p>	<p>当オプションは、フィールド名に関係なく、グループ内のフィールドの内容を別のグループに転送するために使用します。</p> <p>値は定義されたフィールドの順にあるグループから他のグループへフィールドごとに転送されます（再定義で作成されたフィールドは入りません）。</p> <p>個々のフィールドは、どのフォーマットでもかまいません。各グループ内のフィールド数は同じにする必要があります。また、フィールドのレベル構造および配列の次元は一致している必要があります。フォーマット変換は、算術演算の規則に従って行われます。『プログラミングガイド』の「<a href="#">演算割り当てのルール</a>」を参照してください。オペランドにビューを指定することもできます。</p> <p>「<a href="#">例 4 - MOVE BY POSITION</a>」も参照してください。</p>

#### 構文 4 - MOVE EDITED (*operand2* で指定された編集マスク)

```
MOVE EDITED operand1 TO operand2 (EM=value)
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S A	A U B	可	不可
<i>operand2</i>	S A	A U N P I F B D T L	可	可

構文要素の説明：

<b>MOVE EDITED</b>	<p>編集マスクが <i>operand2</i> に対して指定されている場合、<i>operand1</i> の値はこの編集マスクを使用して <i>operand2</i> に配置されます。</p> <p>編集マスクは、<i>operand2</i> の入力編集マスクとみなすことができます。これは、<i>operand1</i> の英数字の内容のどの位置に <i>operand2</i> の大量の入力データが見つかるかを指定するために使用されます。</p> <p>そのため、編集マスクが <i>operand2</i> に存在する以上の文字または桁を参照する場合は、編集マスクは切り捨てられます。<i>operand1</i> の長さは、編集マスクで表される入力値の長さより短くすることはできません。<i>operand1</i> が編集マスクの長さよりも長い場合は、余ったデータが無視されます。</p> <p><i>operand1</i> を編集マスクの長さより長くしないという前提条件のもとでは、次のステートメントを、</p> <pre>MOVE EDITED <i>operand1</i> TO <i>operand2</i> (EM=<i>value</i>)</pre> <p>次のステートメントの実行のような操作とみなすことができます。</p> <pre>STACK TOP DATA <i>operand1</i> INPUT <i>operand2</i> (EM=<i>value</i>)</pre> <p>「<a href="#">例1 - MOVE ステートメントのさまざまな使用例</a>」も参照してください。</p>
<b>EM</b>	<p>編集マスクの詳細については、『パラメータリファレンス』でセッションパラメータ EM を参照してください。</p>

### 構文 5 - MOVE EDITED (operand1 で指定された編集マスク)

```
MOVE EDITED operand1 (EM=value) TO operand2
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S A N	A U N P I F B D T L	可	不可
<i>operand2</i>	S A	A U B	可	可

構文要素の説明：

<b>MOVE EDITED</b>	<p><i>operand1</i> に編集マスクが指定されると、指定の編集マスクを <i>operand1</i> に適用し、結果が <i>operand2</i> に転送されます。</p> <p>編集マスクは、<i>operand1</i> の出力編集マスクとみなすことができます。これは、編集マスクで記述されたレイアウトおよび長さで英数字文字列を作成するために使用されます。<i>operand1</i> のデータ文字または数字の他に、出力文字列に追加の装飾文字を含めることができます。</p> <p>そのため、編集マスクが <i>operand1</i> に存在する以上の文字または桁を参照する場合は、編集マスクは切り捨てられます。作成された出力文字列（編集マスクが適用された後の <i>operand1</i> 値の結果）の長さが <i>operand2</i> の長さを超過しないようにする必要があります。</p> <p><i>operand2</i> を編集マスクの長さより短くしないという前提条件のもとでは、次のステートメントを</p> <pre>MOVE EDITED <i>operand1</i> (EM=<i>value</i>) T0 <i>operand2</i></pre> <p>次のステートメントのような操作とみなすことができます。</p> <pre>WRITE <i>operand1</i> (EM=<i>value</i>)</pre> <p>このステートメントは、画面に出力を表示せずに、変数 <i>operand2</i> に書き込みます。</p> <p>「例 1 - MOVE ステートメントのさまざまな使用例」も参照してください。</p>
<b>EM</b>	<p>編集マスクの詳細については、『パラメータリファレンス』でセッションパラメータ EM を参照してください。</p>

## 構文 6 - MOVE LEFT/RIGHT JUSTIFIED

```
MOVE { LEFT } [JUSTIFIED]operand1 [(parameter)] T0operand2
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S A	N A U N P I F B D T L	可	不可
<i>operand2</i>	S A	A U	可	可

構文要素の説明：

<b>MOVE LEFT/RIGHT JUSTIFIED</b>	当オプションを使用すると、値は <i>operand2</i> に左詰めまたは右詰めで転送されます。 <i>operand2</i> がダイナミック変数のとき、MOVE LEFT/RIGHT JUSTIFIED は使用できません。
<b>MOVE LEFT JUSTIFIED</b>	MOVE LEFT JUSTIFIED で、 <i>operand1</i> のリーディングブランク（空白とバイナリゼロ）は削除されて、 <i>operand2</i> に値が左詰めで置かれます。 <i>operand2</i> の残りは空白で埋められます。値が <i>operand2</i> よりも長いときには、値の右側が切り捨てられます。
<b>MOVE RIGHT JUSTIFIED</b>	MOVE RIGHT JUSTIFIED で、 <i>operand1</i> のトレーリングブランクは切り捨てられて（空白とバイナリゼロが削除されて）、 <i>operand2</i> に値が右詰めで置かれます。 <i>operand2</i> の残りは空白で埋められます。値が <i>operand2</i> よりも長いときには、値の左側が切り捨てられます。  「例1 - MOVE ステートメントのさまざまな使用例」も参照してください。
<i>parameter</i>	MOVE LEFT/RIGHT JUSTIFIED と PM=I を一緒に使用するとき、転送処理は次の手順で行われます。  1. <i>operand1</i> が英数字フォーマットではない場合、値は英数字フォーマットに変換されます。 2. <i>operand1</i> のトレーリングブランク（空白とバイナリゼロ）が削除されます。 3. LEFT JUSTIFIED の場合、 <i>operand1</i> のリーディングブランク（空白とバイナリゼロ）も削除されます。 4. 値を逆に並べ替えてから <i>operand2</i> に移します。 5. 必要に応じて、 <i>operand2</i> の残りを空白で充填するか、または値を切り捨てます（前述参照）。

## 構文 7 - MOVE NORMALIZED

MOVE NORMALIZED ステートメントは、Unicode 文字列を「Unicode Normalization Form C」（NFC）に変換します。結果として生成される Unicode 文字列には、構成前の文字として使用可能な文字の結合順序が含まれません。

ターゲットオペランドの形式が Unicode でない場合は、Unicode からターゲットオペランドへの暗黙的な変換が行われます。この変換中に、デフォルトのコードページ（システム変数「\*CODEPAGE」を参照）が使用されます。

MOVE NORMALIZED ステートメントの詳細は、『Unicode およびコードページのサポート』ドキュメントの「ステートメント」を参照してください。

構文図：

```
MOVE NORMALIZED operand1 TO operand2
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S A	U	可	不可
<i>operand2</i>	S A	A U	可	可

構文要素の説明：

<b>MOVE NORMALIZED</b>	このオプションは、正規化されていない内容を含む可能性のある Unicode フィールドを「Unicode Normalization Form C」(NFC) に変換するために使用されます。この複合形式の Unicode 文字列には、構成前の文字として使用可能な文字の結合順序が含まれません。 <a href="http://www.unicode.org/reports/tr15/#Canonical_Composition_Examples">http://www.unicode.org/reports/tr15/#Canonical_Composition_Examples</a> （「正規化フォーム D と C の例」）も参照してください。  例：  MOVE NORMALIZED #SCR TO #TGT
<i>operand1</i>	変換する Unicode 文字列。
<i>operand2</i>	ターゲットオペランド。

例：

一部のコードポイントは、異なる Unicode 表現を持ちます。例えば、ドイツ語の文字 'Ä' では、Unicode の分解表現は U+0041 に続いて U+0308 であり、結合文字 (U+0308) を使用します。代替表現は、構成前の文字 U+00C4 です。MOVE NORMALIZED ステートメントは、可能な限り構成前の文字を使用して、結合文字を持つ Unicode 表現を正規化された Unicode 表現に変換します。

### 構文 8 - MOVE ENCODED

このセクションでは、MOVE ENCODED ステートメントの構文について説明します。このステートメントの目的の詳細は、『Unicode およびコードページのサポート』ドキュメントの「ステートメント」を参照してください。

構文図：

```
MOVE ENCODED
  operand1 [[IN] CODEPAGE operand2] TO
  operand3 [[IN] CODEPAGE operand4]
  [GIVING operand5]
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
operand1	C S A	A U B	可	不可
operand2	S	A U	可	不可
operand3	S	A U B	可	可
operand4	S A	A U	可	不可
operand5	S	I4	可	可

構文要素の説明：

<b>MOVE ENCODED</b>	MOVE ENCODED ステートメントは、1つのコードページにエンコードされている文字列を、別のコードページの同等の文字列に変換します。
operand1	変換する文字列。
operand2	operand1 のコードページ。operand1 の形式が A または B の場合にのみ指定できます。注 1 および 3 を参照してください。
operand3	ターゲット。  変換結果がターゲットフィールドに適合しない場合は、結果がパディングまたは切り捨てられます。パディング文字には結果のコードページの空白が使用されます。  ダイナミック変数の長さは変換結果の長さに自動的に調整されるため、ターゲットフィールドがダイナミック変数として定義されている場合、パディングや切り捨ては不要です。
operand4	operand3 のコードページ。operand3 の形式が A または B の場合にのみ指定できます。注 1 および 3 を参照してください。
operand5	キーワード GIVING が無い場合は、エラーが発生したときに Natural エラーメッセージが返されます。キーワード GIVING が使用されている場合、operand5 では Natural エラーメッセージの代わりに 0 または Natural エラーコードが返されます。



	ターゲットが切り捨てられた場合、Natural エラーメッセージは返されませんが、キーワード GIVING が使用されている場合は、 <i>operand5</i> に切り捨てを示す適切なエラーコードが格納されます。
--	--

**注意:**

1. コードページオペランドが指定されていない場合は、デフォルトのコードページ（システム変数 \*CODEPAGE の値）が使用されます。
2. ステートメント **SET GLOBALS** またはシステムコマンド GLOBALS のセッションパラメータ CPCVERR が ON に設定されている場合は、ソースフィールドの少なくとも 1 文字を目的のコードページに正しく変換できなかった場合にエラーが出力されますが、ターゲットフィールドでは置換文字で置換されます。
3. ソースモジュール NATCONFIG のマクロ NTCPAGE で定義されているコードページ名のみ使用できます。その他のコードページ名は、対応するランタイムエラーで拒否されます。

例:

```
MOVE ENCODED A-FIELD1 TO A-FIELD2
```

無効：コードページ名がデフォルトで取得され、*operand1* と *operand3* に対して同じであるため、構文エラーになります。

```
MOVE ENCODED A-FIELD1 CODEPAGE 'IBM01140' TO A-FIELD2 CODEPAGE 'IBM01140'
```

無効：コーディングされたコードページ名が *operand1* と *operand3* に対して同じであるため、エラーになります。

```
MOVE ENCODED A-FIELD1 CODEPAGE 'IBM01140' TO A-FIELD2 CODEPAGE 'IBM037'
```

有効：IBM01140 でコーディングされている A-FIELD1 の文字列は、IBM037 でコーディングされた A-FIELD2 に変換されます。

```
MOVE ENCODED U-FIELD TO U-FIELD
```

無効：少なくとも1つのオペランドがAまたはBの形式である必要があるため、エラーになります。

```
MOVE ENCODED U-FIELD TO A-FIELD
```

有効：UTF-16でエンコードされるものとみなされるU-FIELDのUnicode文字列は、デフォルトのコードページ(\*CODEPAGE)の英数字A-FIELDに変換されます。

```
MOVE ENCODED A-FIELD TO U-FIELD
```

有効：デフォルトのコードページ(\*CODEPAGE)でエンコードされるものとみなされるA-FIELDの文字列は、UnicodeフィールドU-FIELDに変換されます。

```
MOVE ENCODED A100-FIELD CODEPAGE 'IBM1140' TO A50-FIELD CODEPAGE 'IBM037'
```

有効：関連するコードページを使用して、A100-FIELD（フォーマット／長さ：A100）からA50-FIELD（フォーマット／長さ：A50）に変換されます。ターゲットは切り捨てられます。Naturalエラーメッセージは返されません。

```
MOVE ENCODED A100-FIELD CODEPAGE 'IBM1140' TO A50-FIELD CODEPAGE 'IBM037' GIVING RC-FIELD
```

有効：関連するコードページを使用して、A100-FIELD（フォーマット／長さ：A100）からA50-FIELD（フォーマット／長さ：A50）に変換されます。ターゲットは切り捨てられます。GIVING節が指定されていないため、RC-FIELDは、値の切り捨てが行われたことを示すエラーコードを受け取ります。

## 例

---

- [例 1 - MOVE ステートメントのさまざまな使用例](#)
- [例 2 - MOVE BY NAME](#)
- [例 3 - 配列を伴う MOVE BY NAME](#)

## ■ 例 4 - MOVE BY POSITION

### 例 1 - MOVE ステートメントのさまざまな使用例

```

** Example 'MOVEX1': MOVE
*****
DEFINE DATA LOCAL
1 #A (N3)
1 #B (A5)
1 #C (A2)
1 #D (A7)
1 #E (N1.0)
1 #F (A5)
1 #G (N3.2)
1 #H (A6)
END-DEFINE
*
MOVE 5 TO #A
WRITE NOTITLE 'MOVE 5 TO #A'          30X '=' #A
*
MOVE 'ABCDE' TO #B #C #D
WRITE 'MOVE ABCDE TO #B #C #D'      20X '=' #B '=' #C '=' #D
*
MOVE -1 TO #E
WRITE 'MOVE -1 TO #E'              28X '=' #E
*
MOVE ROUNDED 1.995 TO #E
WRITE 'MOVE ROUNDED 1.995 TO #E'    18X '=' #E
*
*
MOVE RIGHT JUSTIFIED 'ABC' TO #F
WRITE 'MOVE RIGHT JUSTIFIED ''ABC'' TO #F'      10X '=' #F
*
MOVE EDITED '003.45' TO #G (EM=999.99)
WRITE 'MOVE EDITED ''003.45'' TO #G (EM=999.99)'  4X '=' #G
*
MOVE EDITED 123.45 (EM=999.99) TO #H
WRITE 'MOVE EDITED 123.45 (EM=999.99) TO #H'    6X '=' #H
*
END

```

プログラム **MOVEX1** の出力：

```

MOVE 5 TO #A           #A: 5
MOVE ABCDE TO #B #C #D #E #F #G #H #I #J #K #L #M #N #O #P #Q #R #S #T #U #V #W #X #Y #Z #AA #AB #AC #AD #AE #AF #AG #AH #AI #AJ #AK #AL #AM #AN #AO #AP #AQ #AR #AS #AT #AU #AV #AW #AX #AY #AZ #BA #BB #BC #BD #BE #BF #BG #BH #BI #BJ #BK #BL #BM #BN #BO #BP #BQ #BR #BS #BT #BU #BV #BW #BX #BY #BZ #CA #CB #CC #CD #CE #CF #CG #CH #CI #CJ #CK #CL #CM #CN #CO #CP #CQ #CR #CS #CT #CU #CV #CW #CX #CY #CZ #DA #DB #DC #DD #DE #DF #DG #DH #DI #DJ #DK #DL #DM #DN #DO #DP #DQ #DR #DS #DT #DU #DV #DW #DX #DY #DZ #EA #EB #EC #ED #EE #EF #EG #EH #EI #EJ #EK #EL #EM #EN #EO #EP #EQ #ER #ES #ET #EU #EV #EW #EX #EY #EZ #FA #FB #FC #FD #FE #FF #FG #FH #FI #FJ #FK #FL #FM #FN #FO #FP #FQ #FR #FS #FT #FU #FV #FW #FX #FY #FZ #GA #GB #GC #GD #GE #GF #GG #GH #GI #GJ #GK #GL #GM #GN #GO #GP #GQ #GR #GS #GT #GU #GV #GW #GX #GY #GZ #HA #HB #HC #HD #HE #HF #HG #HH #HI #HJ #HK #HL #HM #HN #HO #HP #HQ #HR #HS #HT #HU #HV #HW #HX #HY #HZ #IA #IB #IC #ID #IE #IF #IG #IH #II #IJ #IK #IL #IM #IN #IO #IP #IQ #IR #IS #IT #IU #IV #IW #IX #IY #IZ #JA #JB #JC #JD #JE #JF #JG #JH #JI #JJ #JK #JL #JM #JN #JO #JP #JQ #JR #JS #JT #JU #JV #JW #JX #JY #JZ #KA #KB #KC #KD #KE #KF #KG #KH #KI #KJ #KK #KL #KM #KN #KO #KP #KQ #KR #KS #KT #KU #KV #KW #KX #KY #KZ #LA #LB #LC #LD #LE #LF #LG #LH #LI #LJ #LK #LL #LM #LN #LO #LP #LQ #LR #LS #LT #LU #LV #LW #LX #LY #LZ #MA #MB #MC #MD #ME #MF #MG #MH #MI #MJ #MK #ML #MM #MN #MO #MP #MQ #MR #MS #MT #MU #MV #MW #MX #MY #MZ #NA #NB #NC #ND #NE #NF #NG #NH #NI #NJ #NK #NL #NM #NO #NP #NQ #NR #NS #NT #NU #NV #NW #NX #NY #NZ #OA #OB #OC #OD #OE #OF #OG #OH #OI #OJ #OK #OL #OM #ON #OO #OP #OQ #OR #OS #OT #OU #OV #OW #OX #OY #OZ #PA #PB #PC #PD #PE #PF #PG #PH #PI #PJ #PK #PL #PM #PN #PO #PP #PQ #PR #PS #PT #PU #PV #PW #PX #PY #PZ #QA #QB #QC #QD #QE #QF #QG #QH #QI #QJ #QK #QL #QM #QN #QO #QP #QQ #QR #QS #QT #QU #QV #QW #QX #QY #QZ #RA #RB #RC #RD #RE #RF #RG #RH #RI #RJ #RK #RL #RM #RN #RO #RP #RQ #RR #RS #RT #RU #RV #RW #RX #RY #RZ #SA #SB #SC #SD #SE #SF #SG #SH #SI #SJ #SK #SL #SM #SN #SO #SP #SQ #SR #SS #ST #SU #SV #SW #SX #SY #SZ #TA #TB #TC #TD #TE #TF #TG #TH #TI #TJ #TK #TL #TM #TN #TO #TP #TQ #TR #TS #TT #TU #TV #TW #TX #TY #TZ #UA #UB #UC #UD #UE #UF #UG #UH #UI #UJ #UK #UL #UM #UN #UO #UP #UQ #UR #US #UT #UU #UV #UW #UX #UY #UZ #VA #VB #VC #VD #VE #VF #VG #VH #VI #VJ #VK #VL #VM #VN #VO #VP #VQ #VR #VS #VT #VU #VV #VW #VX #VY #VZ #WA #WB #WC #WD #WE #WF #WG #WH #WI #WJ #WK #WL #WM #WN #WO #WP #WQ #WR #WS #WT #WU #WV #WW #WX #WY #WZ #XA #XB #XC #XD #XE #XF #XG #XH #XI #XJ #XK #XL #XM #XN #XO #XP #XQ #XR #XS #XT #XU #XV #XW #XX #XY #XZ #YA #YB #YC #YD #YE #YF #YG #YH #YI #YJ #YK #YL #YM #YN #YO #YP #YQ #YR #YS #YT #YU #YV #YW #YX #YY #YZ #ZA #ZB #ZC #ZD #ZE #ZF #ZG #ZH #ZI #ZJ #ZK #ZL #ZM #ZN #ZO #ZP #ZQ #ZR #ZS #ZT #ZU #ZV #ZW #ZX #ZY #ZZ

```

## 例 2 - MOVE BY NAME

```

** Example 'MOVEX2': MOVE BY NAME
*****
DEFINE DATA LOCAL
1 #SBLOCK
  2 #FIELD1 (A10) INIT <'AAAAAAAAAA'>
  2 #FIELD2 (A10) INIT <'BBBBBBBBBB'>
  2 #FIELD3 (A10) INIT <'CCCCCCCC'>
  2 #FIELD4 (A10) INIT <'DDDDDDDD'>
1 #TBLOCK
  2 #FIELD1 (A15) INIT <' '>
  2 #FIELD2 (A10) INIT <' '>
  2 #FIELD3 (A10) INIT <' '>
  2 #FIELD4 (A10) INIT <' '>
  2 #FIELD5 (A20) INIT <' '>
  2 #FIELD6 (A10) INIT <' '>
END-DEFINE
*
MOVE BY NAME #SBLOCK TO #TBLOCK
*
WRITE NOTITLE 'CONTENTS OF #TBLOCK AFTER MOVE BY NAME:'
  // '=' #TBLOCK.#FIELD1
  / '=' #TBLOCK.#FIELD2
  / '=' #TBLOCK.#FIELD3
  / '=' #TBLOCK.#FIELD4
  / '=' #TBLOCK.#FIELD5
  / '=' #TBLOCK.#FIELD6
*
END

```

**MOVE BY NAME** 処理後の #TBLOCK の内容：

```
CONTENTS OF #TBLOCK AFTER MOVE BY NAME:
```

```
#FIELD1:
#FIELD A: AAAAAAAAAA
#FIELD2:
#FIELD B: BBBBBBBBBB
#FIELD3:
#FIELD C: CCCCCCCCCC
```

**例 3 - 配列を伴う MOVE BY NAME**

```
DEFINE DATA LOCAL
  1 #GROUP1
    2 #FIELD (A10/1:10)
  1 #GROUP2
    2 #FIELD (A10/1:10)
END-DEFINE
...
MOVE BY NAME #GROUP1 TO #GROUP2
...
```

この例では、MOVE ステートメントは内部的に次のように解決されます。

```
MOVE #GROUP1.#FIELD (*) TO #GROUP2.#FIELD (*)
```

添字付きグループの一部が同一グループの他の部分に転送される場合、次の例のような予期しない結果となります。

```
DEFINE DATA LOCAL
  1 #GROUP1 (1:5)
    2 #FIELD A (N1) INIT <1,2,3,4,5>
    2 REDEFINE #FIELD A
      3 #FIELD B (N1)
END-DEFINE
...
MOVE BY NAME #GROUP1 (2:4) TO #GROUP1 (1:3)
...
```

この例では、MOVE ステートメントは内部的に次のように解決されます。

```
MOVE #FIELDA (2:4) TO #FIELDA (1:3) MOVE #FIELDB (2:4) TO #FIELDB (1:3)
```

まず、#FIELDA のオカレンス 2 から 4 の内容は、#FIELDA のオカレンス 1 から 3 に移動されます。つまり、オカレンスは次の値を受け取ります。

オカレンス	1.	2.	3.	4.	5.
前の値：	1	2	3	4	5
後の値：	2	3	4	4	5

#FIELDB のオカレンス 2 から 4 の内容は、#FIELDB のオカレンス 1 から 3 に移動されます。つまり、オカレンスは次の値を受け取ります。

オカレンス	1.	2.	3.	4.	5.
前の値：	2	3	4	4	5
後の値：	3	4	4	4	5

#### 例 4 - MOVE BY POSITION

```
DEFINE DATA LOCAL
  1 #GROUP1
    2 #FIELD1A (N5)
    2 #FIELD1B (A3/1:3)
    2 REDEFINE #FIELD1B
      3 #FIELD1BR (A9)
  1 #GROUP2
    2 #FIELD2A (N5)
    2 #FIELD2B (A3/1:3)
    2 REDEFINE #FIELD2B
      3 #FIELD2BR (A9)
END-DEFINE
...
MOVE BY POSITION #GROUP1 TO #GROUP2
...
```

#FIELD1A の内容は、#FIELD2A に転送され、#FIELD1B の内容は、#FIELD2B に転送されます。フィールド #FIELD1BR と #FIELD2BR は影響を受けません。

# 83 MOVE ALL

---

■ 機能 .....	566
■ 構文説明 .....	566
■ 例 .....	567

```
MOVE ALL operand1 TO operand2 [UNTIL operand3]
```

関連ステートメント：**ADD** | **COMPRESS** | **COMPUTE** | **DIVIDE** | **EXAMINE** | **MOVE** | **MULTIPLY** | **RESET** | **SEPARATE** | **SUBTRACT**

関連機能グループ：「[算術演算とデータ移動操作](#)」

## 機能

MOVE ALL ステートメントは、*operand1* の値を *operand2* がいっぱいになるまで *operand2* に繰り返し移動するために使用します。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A U N B	可	不可
<i>operand2</i>	S A	A U B	可	可
<i>operand3</i>	C S	N P I	可	不可

構文要素の説明：

<i>operand1</i>	<p>ソースオペランド：</p> <p>ソースオペランドには転送する値が含まれています。</p> <p>リーディングゼロを含む数値オペランドのすべての桁が転送されます。</p>
TO <i>operand2</i>	<p>ターゲットオペランド：</p> <p>ターゲットオペランドは、MOVE ALL 処理の前にリセットされません。このことは UNTIL オプションを使用するとき特に重要です。<i>operand2</i> にあらかじめ存在していたデータは、MOVE ALL 処理で明確に上書きしないと、そのまま残されるからです。</p>
UNTIL <i>operand3</i>	<p>UNTIL オプション：</p> <p>UNTIL オプションは <i>operand2</i> に対する桁数を指定することにより MOVE ALL 処理を制限するために使用します。<i>operand3</i> には桁数を指定します。MOVE ALL 処理はこの値に達したときに終了します。</p> <p><i>operand3</i> が <i>operand2</i> の長さより大きい場合、MOVE ALL 処理は <i>operand2</i> がいっぱいになったときに終了します。</p>



UNTIL オプションを使用して、ダイナミック変数に初期値を割り当てることもできます。*operand2* がダイナミック変数のとき、MOVE ALL 操作後のダイナミック変数の長さは、*operand3* の値に対応します。ダイナミック変数の現在の長さは、システム変数 \*LENGTH を使用して確認できます。ダイナミック変数については、「ダイナミック変数の使用」を参照してください。

## 例

```
** Example 'MOAEX1': MOVE ALL
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 FIRST-NAME
  2 NAME
  2 CITY
1 VEH-VIEW VIEW OF VEHICLES
  2 PERSONNEL-ID
  2 MAKE
END-DEFINE
*
LIMIT 4
RD. READ EMPLOY-VIEW BY NAME
  SUSPEND IDENTICAL SUPPRESS
  /*
  FD. FIND VEH-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (RD.)
    IF NO RECORDS FOUND
      MOVE ALL '*' TO FIRST-NAME (RD.)
      MOVE ALL '*' TO CITY (RD.)
      MOVE ALL '*' TO MAKE (FD.)
    END-NOREC
  /*
  DISPLAY NOTITLE (ES=OFF IS=ON ZP=ON AL=15)
    NAME (RD.) FIRST-NAME (RD.)
    CITY (RD.)
    MAKE (FD.) (IS=OFF)
  /*
END-FIND
END-READ
END
```


プログラム **MOAEX1** の出力：

NAME	FIRST-NAME	CITY	MAKE
ABELLAN	*****	*****	*****
ACHIESON	ROBERT	DERBY	FORD
ADAM	*****	*****	*****
ADKINSON	JEFF	BROOKLYN	GENERAL MOTORS

# 84 MOVE INDEXED

---

MOVE INDEXED ステートメントは、互換性保持のためにのみサポートされています。

 **注意:** 配列オペランドを伴う MOVE ステートメントとは対照的に、MOVE INDEXED ステートメントが実行される場合に範囲外の添字値はチェックできません。そのため、不正な MOVE INDEXED ステートメントを実行すると、ユーザーデータが不用意に破壊されることがあります。

したがって、MOVE INDEXED ステートメントを MOVE ステートメントで置き換えることを強くお勧めします。

詳細については、「[MOVE](#)」ステートメントを参照してください。



# 85 MULTIPLY

---

■ 機能 .....	572
■ 構文説明 .....	572
■ 例 .....	574

関連ステートメント：[ADD](#) | [COMPRESS](#) | [COMPUTE](#) | [DIVIDE](#) | [EXAMINE](#) | [MOVE](#) | [MOVE ALL](#) | [RESET](#) | [SEPARATE](#) | [SUBTRACT](#)

関連機能グループ：「[算術演算とデータ移動操作](#)」

## 機能

---

MULTIPLY ステートメントは、2つのオペランドを乗算するために使用します。使用される構文に応じて、乗算結果は *operand1* または *operand3* に入ります。

データベースフィールドを結果フィールドとして使用するとき、MULTIPLY 演算は、プログラム内で使用したフィールドの内部値だけを更新します。データベース内のフィールドの値は変更されないまま維持されます。

配列を含む乗算については、『プログラミングガイド』の「[算術演算の規則](#)」、「[配列での算術演算](#)」も参照してください。

## 構文説明

---

このステートメントには、2つの異なる構造が可能です。

- [構文 1 - GIVING 節のない MULTIPLY](#)
- [構文 2 - GIVING 節のある MULTIPLY](#)

### 構文 1 - GIVING 節のない MULTIPLY

構文 1 が使用される場合、乗算の結果は *operand1* に入ります。

```
MULTIPLY [ROUNDED] operand1 BY operand2
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

オペランド定義テーブル（構文 1）：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A M	N P I F	可	不可
<i>operand2</i>	C S A N	N P I F	可	不可

構文要素の説明（構文 1）

<i>operand1</i> BY <i>operand2</i>	<i>operand1</i> は被乗数、 <i>operand2</i> は乗数です。GIVING節が使用されていない場合は、 <i>operand1</i> に結果が入るため、ステートメントは次と同等になります。  <pre>&lt;oper1&gt; := &lt;oper1&gt; * &lt;oper2&gt;</pre>
ROUNDED	キーワード ROUNDED を指定すると、値は <i>operand1</i> または <i>operand3</i> に割り当てられる前に切り上げられます。切り上げについては、『プログラミングガイド』の「演算割り当てのルール」、「フィールドの切り捨てと切り上げ」を参照してください。

## 構文 2 - GIVING 節のある MULTIPLY

構文 2 が使用される場合、乗算の結果は *operand3* に入ります。

```
MULTIPLY [ROUNDED] operand1 BY operand2 GIVING operand3
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

オペランド定義テーブル（構文 2）：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S A M	N P I F	可	不可
<i>operand2</i>	C S A N	N P I F	可	不可
<i>operand3</i>	S A M A U N P I F B*	T	可	可

\**operand3* のフォーマット B は、4 以下の長さでのみ使用できます。

## 構文要素の説明（構文 2）

<i>operand1</i> <b>BY</b>	<i>operand1</i> は被乗数、 <i>operand2</i> は乗数です。GIVING節が使用されている場合は、 <i>operand1</i> が変更されず、結果が <i>operand3</i> に入るため、ステートメントは次と同等になります。
<i>operand2</i> <b>GIVING</b>	
<i>operand3</i>	<pre>&lt;oper3&gt; := &lt;oper1&gt; * &lt;oper2&gt;</pre> <p><i>operand1</i>が数値定数の場合は、GIVING節が必要です。</p>
<b>ROUNDED</b>	キーワード <b>ROUNDED</b> を指定すると、値は <i>operand1</i> または <i>operand3</i> に割り当てられる前に切り上げられます。切り上げについては、『プログラミングガイド』の「演算割り当てのルール」、「フィールドの切り捨てと切り上げ」を参照してください。

## 例

```

** Example 'MULEX1': MULTIPLY
*****
DEFINE DATA LOCAL
1 #A      (N3) INIT <20>
1 #B      (N5)
1 #C      (N3.1)
1 #D      (N2)
1 #ARRAY1 (N5/1:4,1:4) INIT (2,*) <5>
1 #ARRAY2 (N5/1:4,1:4) INIT (4,*) <10>
END-DEFINE
*
MULTIPLY #A BY 3
WRITE NOTITLE 'MULTIPLY #A BY 3'          25X '=' #A
*
MULTIPLY #A BY 3 GIVING #B
WRITE 'MULTIPLY #A BY 3 GIVING #B'      15X '=' #B
*
MULTIPLY ROUNDED 3 BY 3.5 GIVING #C
WRITE 'MULTIPLY ROUNDED 3 BY 3.5 GIVING #C' 6X '=' #C
*
MULTIPLY 3 BY -4 GIVING #D
WRITE 'MULTIPLY 3 BY -4 GIVING #D'      14X '=' #D
*
MULTIPLY -3 BY -4 GIVING #D
WRITE 'MULTIPLY -3 BY -4 GIVING #D'     14X '=' #D
*
MULTIPLY 3 BY 0 GIVING #D
WRITE 'MULTIPLY 3 BY 0 GIVING #D'       14X '=' #D
*
WRITE / '=' #ARRAY1 (2,*) '=' #ARRAY2 (4,*)
MULTIPLY #ARRAY1 (2,*) BY #ARRAY2 (4,*)

```



```
WRITE / 'MULTIPLY #ARRAY1 (2,*) BY #ARRAY2 (4,*)'  
      / '=' #ARRAY1 (2,*) '=' #ARRAY2 (4,*)  
*  
END
```

プログラム **MULEX1** の出力：

```
MULTIPLY #A BY 3 #A: 60  
MULTIPLY #A BY 3 GIVING #B #B: 180  
MULTIPLY ROUNDED 3 BY 3.5 GIVING #C #C: 10.5  
MULTIPLY 3 BY -4 GIVING #D #D: -12  
MULTIPLY -3 BY -4 GIVING #D #D: 12  
MULTIPLY 3 BY 0 GIVING #D #D: 0  
  
#ARRAY1: 5 5 5 5 #ARRAY2: 10 10 10 10  
  
MULTIPLY #ARRAY1 (2,*) BY #ARRAY2 (4,*)  
#ARRAY1: 50 50 50 50 #ARRAY2: 10 10 10 10
```



# 86 NEWPAGE

---

■ 機能 .....	578
■ 構文説明 .....	578
■ 例 .....	579

NEWPAGE [( <i>rep</i> )]	[	{	EVEN [IF] TOP [OF] [PAGE]	}	[	{	IF	}	WHEN	]	LESS [THAN] <i>operand1</i> [LINES] [LEFT]	]	]
													[[WITH] TITLE <i>title-definition</i> ]

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT END OF PAGE](#) | [AT TOP OF PAGE](#) | [CLOSE PRINTER](#) | [DEFINE PRINTER](#) | [DISPLAY](#) | [EJECT](#) | [FORMAT](#) | [PRINT](#) | [SKIP](#) | [SUSPEND IDENTICAL SUPPRESS](#) | [WRITE](#) | [WRITE TITLE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

NEWPAGE ステートメントは、ページ送りをするために使用します。また、NEWPAGE では [AT END OF PAGE](#) および [WRITE TRAILER](#) ステートメントも実行されます。[WRITE TITLE](#)、[WRITE NOTITLE](#) あるいは [DISPLAY NOTITLE](#) ステートメントを指定して固有のタイトル処理を定義していない限り、日付、時刻、ページ番号を含むデフォルトのタイトルが新しいページごとに出力されます。



### 注意:

1. ページ送りは、NEWPAGE ステートメントが実行されるときに行われるわけではありません。出力を生成する後続のステートメントが実行されるときにのみ行われます。
2. NEWPAGE ステートメントを使用しないとき、ページ送りは Natural セッションパラメータ PS に基づいて自動的に制御されます。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	N P I	可	不可

## 構文要素の説明：

<i>(rep)</i>	<p>レポート指定：</p> <p>表記 (<i>rep</i>) は、NEWPAGE ステートメントを適用するレポートの ID を指定するために使用できます。</p> <p>範囲 0~31 の値、または <b>DEFINE PRINTER</b> ステートメントを使用して割り当てた論理名を指定できます。</p> <p>(<i>rep</i>) を指定しない場合、NEWPAGE ステートメントは最初のレポート (レポート 0) に適用されます。</p> <p>Natural で作成される出力レポートの形式を制御する方法については、『プログラミングガイド』の「データ出力制御」を参照してください。</p>
<b>EVEN IF TOP OF PAGE</b>	<p>当オプションは、NEWPAGE ステートメントの実行の直前にページ送りが行われていても、新しいページを (対応する <b>AT TOP OF PAGE</b> およびページタイトル処理とともに) 生成します。</p>
<b>WHEN LESS THAN</b> <i>operand1 LINES LEFT</i>	<p>当オプションは、現ページの残りの行数が <i>operand1</i> で指定された行数よりも少ないとき、ページ換えをするために使用します (現在の行はセッションパラメータ PS の値と比較されます)。</p>
<b>WITH TITLE</b> <i>title-definition</i>	<p><b>WITH TITLE</b> オプションは、新しいページに変わったときのタイトルを指定するために使用します。 <i>title-definition</i> の指定に使用する構文は、NEWPAGE <b>WITH TITLE</b> <i>title-definition</i> ステートメントでは <b>SKIP</b> 節が使用できないことを除けば、<b>WRITE TITLE</b> ステートメントの構文と同じです。</p>

## 例

```

** Example 'NWPEX1': NEWPAGE
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 NAME
  2 SALARY      (1)
  2 CURR-CODE (1)
END-DEFINE
*
LIMIT 15
READ EMPLOY-VIEW BY CITY FROM 'DENVER'
  DISPLAY CITY (IS=ON) NAME SALARY (1) CURR-CODE (1)
  AT BREAK OF CITY
  SKIP 1
  /*
  NEWPAGE WHEN LESS THAN 10 LINES LEFT
  WRITE '*****'
  / 'SUMMARY FOR ' OLD(CITY)

```

```

/ '*****'
/ '*****'
/ 'SUM OF SALARIES:' SUM(SALARY(1))
/ 'AVG OF SALARIES:' AVER(SALARY(1))
/ '*****'
NEWPAGE
/*
END-BREAK
END-READ
END

```

プログラム **NWPEX1** の出力 - ページ 1 :

Page 1 05-01-18 10:01:45

CITY	NAME	ANNUAL SALARY	CURRENCY CODE
DENVER	TANIMOTO	33000	USD
	MEYER	50000	USD

\*\*\*\*\*  
SUMMARY FOR DENVER  
\*\*\*\*\*  
\*\*\*\*\*  
SUM OF SALARIES: 83000  
AVG OF SALARIES: 41500  
\*\*\*\*\*

プログラム **NWPEX1** の出力 - ページ 2 :

Page 2 05-01-18 10:01:45

CITY	NAME	ANNUAL SALARY	CURRENCY CODE
DERBY	DEAKIN	8750	UKL
	GARFIELD	6750	UKL
	MUNN	8800	UKL
	MUNN	5650	UKL
	GREBBY	9550	UKL
	WHITT	8650	UKL
	PONSONBY	5500	UKL
	MAGUIRE	4150	UKL
	HEYWOOD	3900	UKL
	BRYDEN	6750	UKL
	SMITH	39000	UKL
	CONQUEST	45000	UKL
	ACHIESON	11300	UKL

```
*****  
SUMMARY FOR DERBY  
*****
```

プログラム NWPEX1 の出力 - ページ 3 :

```
-----  
DERBY          DEAKIN          8750 UKL  
                GARFIELD        6750 UKL  
                MUNN           8800 UKL  
                MUNN           5650 UKL  
                GREBBY        9550 UKL  
                WHITT         8650 UKL  
                PONSONBY      5500 UKL  
                MAGUIRE       4150 UKL  
                HEYWOOD       3900 UKL  
                BRYDEN        6750 UKL  
                SMITH         39000 UKL  
                CONQUEST      45000 UKL  
                ACHIESON      11300 UKL
```

```
*****  
SUMMARY FOR DERBY  
*****  
*****  
SUM OF SALARIES:      163750  
AVG OF SALARIES:      12596  
*****
```





# 87

## OBTAIN

---

▪ 機能 .....	584
▪ 制限事項 .....	584
▪ 構文説明 .....	585
▪ 例 .....	589

```
OBTAIN operand1 ...
```

## 機能

---

OBTAIN ステートメントは、レポートモードで1つ以上のフィールドをファイルから読み取るときに使用します。OBTAIN ステートメントは、Natural オブジェクトプログラムの実行コードを生成するものではありません。主に、マルチプルバリュースフィールドの値範囲またはピリオディックグループのオカレンスの範囲を読み取って、これらの範囲の一部を後でプログラムで参照できるようにするために使用されます。

Natural は後続のステートメント（DISPLAY ステートメントや COMPUTE ステートメントなど）で参照される各データベースフィールドを自動的に読み取るため、プログラムで参照するデータベースフィールドごとに OBTAIN ステートメントを使用する必要はありません。

配列形式のマルチプルバリュースフィールドまたはピリオディックグループフィールドが参照されている場合は、配列を OBTAIN ステートメントで定義して、フィールドのすべてのオカレンスに対して作成されるようにする必要があります。配列が定義される前に個々のマルチプルバリュースフィールドまたはピリオディックグループフィールドが参照された場合は、フィールドが配列に配置されず、配列からは独立して存在するようになります。フィールドには、配列内の対応するオカレンスと同じ値が含まれます。

2つ目の個々のオカレンスまたは配列の配列次元が初期配列に含まれている場合、マルチプルバリュースフィールド、ピリオディックグループフィールドまたはサブ配列の個々のオカレンスは、以前に定義した配列に保持できます。

一意の変数添字のあるマルチプルバリュースフィールドまたはピリオディックグループフィールドへの参照は、値の配列に含めることができません。配列の個々のオカレンスを変数添字とともに処理する場合は、添字表現の接頭辞として一意の変数添字を付けて、個々の配列を示す必要があります。

## 制限事項

---

OBTAIN ステートメントはレポートモードでだけ有効です。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	SAG	AUNPIFBDTL	可	可

構文要素の説明：

<i>operand1</i>	<i>operand1</i> で、OBTAIN ステートメントの結果、参照可能になるフィールドを指定します。
-----------------	---

例：

```
READ FINANCE OBTAIN CREDIT-CARD (1-10)
DISPLAY CREDIT-CARD (3-5) CREDIT-CARD (6-8)
SKIP 1 END
```

上記の例では、フィールド CREDIT-CARD（ピリオディックグループに含まれます）の最初の 10 個のオカレンスが読み取られ、オカレンス 3~5 と 6~8 が表示されます。後続のサブ配列は初期配列（1~10）に存在するようになります。

```
READ FINANCE
MOVE 'ONE' TO CREDIT-CARD (1)
DISPLAY CREDIT-CARD (1) CREDIT-CARD (1-5)
```

出力：

```

CREDIT-CARD      CREDIT-CARD
-----
ONE              DINERS CLUB
                AMERICAN EXPRESS

ONE              AVIS
                AMERICAN EXPRESS

ONE              HERTZ
                AMERICAN EXPRESS
```

```
ONE                UNITED AIR TRAVEL
```

CREDIT-CARD (1) への最初の参照は、配列に含まれていません。一意のオカレンス (1) への参照の後に定義されている配列は、一意のオカレンスまたは定義されている配列より短い配列を避及的に含むことはできません。

```
READ FINANCE
OBTAIN CREDIT-CARD (1-5)
MOVE 'ONE' TO CREDIT-CARD (1)
DISPLAY CREDIT-CARD (1) CREDIT-CARD (1-5)
```

出力：

CREDIT-CARD	CREDIT-CARD
ONE	ONE AMERICAN EXPRESS
ONE	ONE AMERICAN EXPRESS
ONE	ONE AMERICAN EXPRESS
ONE	ONE

CREDIT-CARD (1) への個々の参照は、OBTAIN ステートメントで定義されている配列に含まれません。

```
MOVE (1) TO INDEX
READ FINANCE
DISPLAY CREDIT-CARD (1-5) CREDIT-CARD (INDEX)
```

出力：

CREDIT-CARD	CREDIT-CARD
DINERS CLUB AMERICAN EXPRESS	DINERS CLUB
AVIS AMERICAN EXPRESS	AVIS
HERTZ AMERICAN EXPRESS	HERTZ
UNITED AIR TRAVEL	UNITED AIR TRAVEL

変数添字表記を使用した CREDIT-CARD への参照は、配列に含まれません。

```

RESET A(A20) B(A20) C(A20)
MOVE 2 TO I (N3)
MOVE 3 TO J (N3)
READ FINANCE
  OBTAIN CREDIT-CARD (1:3) CREDIT-CARD (I:I+2) CREDIT-CARD (J:J+2)
  FOR K (N3) = 1 TO 3
    MOVE CREDIT-CARD (1.K) TO A
    MOVE CREDIT-CARD (I.K) TO B
    MOVE CREDIT-CARD (J.K) TO C
  DISPLAY A B C
  LOOP /* FOR
LOOP / * READ
END

```

出力：

A	B	C
CARD 01	CARD 02	CARD 03
CARD 02	CARD 03	CARD 04
CARD 03	CARD 04	CARD 05

3つの配列は、一意の基本添字を添字表現の修飾子として使用して、個別にアクセスできます。

### 無効な例 1

```
READ FINANCE
OBTAIN CREDIT-CARD (1-10)
FOR I 1 10
MOVE CREDIT-CARD (I) TO A(A20)
WRITE A
END
```

上記の例では、レコードの読み取り時（READ）に添字 I にまだ値 0 が含まれているため、エラーメッセージ NAT1006（変数添字の値 = 0）が生成されます。

いずれの場合にも、上記の例では、変数添字を持つ個々のオカレンスを配列に含めることができず、変数添字（I）は次のレコードの読み取り時にのみ評価されるため、CREDIT-CARD の最初の 10 個のオカレンスは出力されません。

上記を実行する正しいメソッドは次のとおりです。

```
READ FINANCE
OBTAIN CREDIT-CARD (1-10)
FOR I 1 10
MOVE CREDIT-CARD (1.I) TO A (A20)
WRITE A
END
```

### 無効な例 2

```
READ FINANCE
FOR I 1 10
WRITE CREDIT-CARD (I)
END
```

上記の例では、レコードが READ ステートメントで読み取られたときに添字 I がゼロであるため、エラーメッセージ NAT1006 が生成されます。

上記を実行する正しいメソッドは次のとおりです。

```
READ FINANCE
FOR I 1 10
GET SAME
WRITE CREDIT-CARD (0030/I)
END
```

GET SAME ステートメントは、変数添字が FOR ループ内で更新された後にレコードを再び読み取るために必要です。

## 例

- 例 1 - OBTAIN ステートメント
- 例 2 - 複数の範囲がある OBTAIN ステートメント

### 例 1 - OBTAIN ステートメント

```

** Example 'OBTEX1': OBTAIN
*****
RESET #INDEX (I1)
*
LIMIT 5
READ EMPLOYEES BY CITY
  OBTAIN SALARY (1:4)
  /*
  IF SALARY (4) GT 0 DO
    WRITE '=' NAME / 'SALARIES (1:4):' SALARY (1:4)
    FOR #INDEX 1 TO 4
      WRITE 'SALARY' #INDEX SALARY (1.#INDEX)
    LOOP
    SKIP 1
  DOEND
LOOP
*
END

```

プログラム **OBTEX1** の出力：

```

Page      1                                05-02-08  13:37:48

NAME: SENKO
SALARIES (1:4):      31500      29900      28100      26600
SALARY   1      31500
SALARY   2      29900
SALARY   3      28100
SALARY   4      26600

NAME: HAMMOND
SALARIES (1:4):      22000      20200      18700      17500
SALARY   1      22000
SALARY   2      20200
SALARY   3      18700
SALARY   4      17500

```

## 例 2 - 複数の範囲がある OBTAIN ステートメント

```

** Example 'OBTEX2': OBTAIN (with multiple ranges)
*****
RESET #INDEX (I1) #K (I1)
*
#INDEX := 2
#K := 3
*
LIMIT 2
*
READ EMPLOYEES BY CITY
  OBTAIN SALARY (1:5)
    SALARY (#INDEX:#INDEX+3)
/*
IF SALARY (5) GT 0 DO
  WRITE '=' NAME
  WRITE 'SALARIES (1-5):' SALARY (1:5) /
  WRITE 'SALARIES (2-5):' SALARY (#INDEX:#INDEX+3)
  WRITE 'SALARIES (2-5):' SALARY (#INDEX.1:4) /
  WRITE 'SALARY 3:' SALARY (3)
  WRITE 'SALARY 3:' SALARY (#K)
  WRITE 'SALARY 4:' SALARY (#INDEX.#K)
DOEND
LOOP

```

プログラム **OBTEX2** の出力：

```

Page      1                                05-02-08  13:38:31

NAME: SENKO
SALARIES (1-5):      31500      29900      28100      26600      25200

SALARIES (2-5):      29900      28100      26600      25200
SALARIES (2-5):      29900      28100      26600      25200

SALARY 3:           28100
SALARY 3:           28100
SALARY 4:           26600

```

OBTAIN ステートメントの詳細な使用例については、を参照してください。

「データベース配列の参照」（『プログラミングガイド』）



# 88 ON ERROR

---

■ 機能 .....	592
■ 制限事項 .....	592
■ 構文説明 .....	593
■ サブルーチン内の ON ERROR 処理 .....	593
■ システム変数 *ERROR-NR および *ERROR-LINE .....	593
■ 例 .....	594

ストラクチャードモード構文

```
ON ERROR  
  statement ...  
END-ERROR
```

レポートイングモード構文

```
ON ERROR { statement...  
          DO statement ...DOEND }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：**DECIDE FOR** | **DECIDE ON** | **IF** | **IF SELECTION**

関連機能グループ：「[論理条件の処理](#)」

## 機能

---

ON ERROR ステートメントは実行時のエラーを拾い、Natural エラーメッセージを別の形で出力し、Natural プログラムの実行を終了させ、コマンド入力モードに戻すために使用します。

ON ERROR ステートメントブロックの実行が開始されると、プログラムの正常な流れは中断され、再開することはできません。ただしエラー 3145（要求したレコードがホールド状態にある）の場合は、RETRY ステートメントによって処理が中断された所から再開されます。

このステートメントは非手続き型なので、プログラム内の位置ではなくイベントによって実行されます。

## 制限事項

---

各 Natural オブジェクト内で 1 回だけ、ON ERROR ステートメントを使用できます。

## 構文説明

<i>statement...</i>	<p><b>ON ERROR</b> 処理の定義：</p> <p>ON ERROR 条件が発生したときに実行される処理を定義するには、1つまたは複数のステートメントを指定できます。</p> <p><b>ON ERROR</b> ブロックからの抜け出し</p> <p><b>FETCH</b>、<b>STOP</b>、<b>TERMINATE</b>、<b>RETRY</b> または <b>ESCAPE ROUTINE</b> の各ステートメントを使用して、ON ERROR ブロックから抜け出すことができます。これらのステートメントのいずれかを使用して抜け出さないと、通常のエラーメッセージ処理が行われ、プログラムの実行が終了します。</p>
<b>END-ERROR</b>	Natural 予約語 <b>END-ERROR</b> を使用して、ON ERROR ステートメントブロックを終了させる必要があります。

## サブルーチン内の ON ERROR 処理

**CALLNAT**、**PERFORM** または **FETCH RETURN** を使用したサブルーチン構成になっている場合、各モジュールに ON ERROR ステートメントが指定可能です。

エラーが起きた場合、Natural は自動的にサブルーチン構成に従って戻り、実行中のサブルーチン内で最初に見つかった ON ERROR ステートメントを選択します。ON ERROR ステートメントがどのレベルのどのモジュールにもない場合は、通常のエラーメッセージ処理を行い、プログラムの実行が終了します。

## システム変数 \*ERROR-NR および \*ERROR-LINE

次の Natural システム変数は、（次の例で示すように）ON ERROR ステートメントと組み合わせて使用できます。

<b>*ERROR-NR</b>	Natural で検出されたエラーの数が含まれます。
<b>*ERROR-LINE</b>	エラーの原因となったステートメントの行番号が含まれます。

## 例

```
** Example 'ONEEX1': ON ERROR
**
**
CAUTION: Executing this example will modify the database records!
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CITY
*
1 #NAME (A20)
1 #CITY (A20)
END-DEFINE
*
REPEAT
  INPUT 'ENTER NAME:' #NAME
  IF #NAME = ' '
    STOP
  END-IF
  FIND EMPLOY-VIEW WITH NAME = #NAME
  INPUT (AD=M) 'ENTER NEW VALUES:' ///
    'NAME:' NAME /
    'CITY:' CITY

  UPDATE
  END TRANSACTION
/*
ON ERROR
  IF *ERROR-NR = 3009
    WRITE 'LAST TRANSACTION NOT SUCCESSFUL'
      / 'HIT ENTER TO RESTART PROGRAM'
    FETCH 'ONEEX1'
  END-IF
  WRITE 'ERROR' *ERROR-NR 'OCCURRED IN PROGRAM' *PROGRAM
    'AT LINE' *ERROR-LINE
  FETCH 'MENU'
END-ERROR
/*
END-FIND
END-REPEAT
END
```

# 89

## OPEN CONVERSATION

---

■ 機能 .....	596
■ 構文説明 .....	596
■ 詳細と例 .....	597

```
OPEN CONVERSATION USING [SUBPROGRAMS] {operand1} ...
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CLOSE CONVERSATION](#) | [DEFINE DATA CONTEXT](#)

関連機能グループ：「[Natural リモートプロシージャコール](#)」

## 機能

OPEN CONVERSATION ステートメントは、Natural リモートプロシージャコール (RPC) と一緒に使用します。これにより、クライアントは会話を開き、会話に取り込むリモートサブプログラムを指定することができます。

OPEN CONVERSATION ステートメントを実行すると、会話を識別するユニーク ID がシステム変数 \*CONVID に割り当てられます。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
operand1	C S A	A	可	不可

構文要素の説明：

operand1	<p>サブプログラム名：</p> <p>operand1 として、会話に取り込むリモートサブプログラムの名前を指定します。</p> <p>サブプログラム名は1~8文字の定数または1~8桁の英数字変数で指定することができます。</p>
----------	--

## 詳細と例

---

『*Natural* リモートプロシージャコール (RPC) 』ドキュメントの次のセクションを参照してください。

- 会話型モードでの *Natural* RPC の動作
- 会話型 RPC の使用





# 90      OPTIONS

---

▪ 機能 .....	600
▪ 複数の OPTIONS ステートメントの処理 .....	600

`OPTIONS parameter ...`

## 機能

---

OPTIONS ステートメントを使用すると、現在の Natural プログラミングオブジェクトに、コンパイルオプションをパラメータとして指定できます。これらは、以下のように指定できるオプションと同じです。

- NTCMPO マクロを使用して静的に指定
- CMPO パラメータを使用して動的に指定
- COMPOPT システムコマンドを使用して Natural セッション内で指定

また、OPTIONS ステートメントを使用して、Natural Optimizer Compiler のオプションを指定できます。これらのオプションについては、『*Natural Optimizer Compiler*』のドキュメントで説明されています。

MCG オプションで指定された Natural Optimizer Compiler オプションは、Natural Optimizer Compiler がインストールされていない場合にも有効性が検証されます。

## 複数の OPTIONS ステートメントの処理

---

複数の OPTIONS ステートメントが同じプログラミングオブジェクトに指定されている場合、オプション設定は即時に有効になります。ただし、PSIGNF、TSENABL、GFID の各オプションには、このことは当てはまりません。これらのオプションの場合は、最後の OPTIONS ステートメントで指定されたオプション値が適用されます。

# 91 PARSE XML

---

▪ 機能 .....	602
▪ 構文説明 .....	603
▪ 例 .....	605

```

PARSE XML operand1 [INTO [PATH operand2] [NAME operand3] [VALUE operand4]]
  [NORMALIZE]NAMESPACE operand5PREFIX operand6]
  statement...
END-PARSE (structured mode only)
[LOOP] (reporting mode only)
    
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[インターネットとXML](#)」

## 機能

PARSE XML ステートメントを使用すると、Natural プログラムから XML ドキュメントを解析できるようになります。『プログラミングガイド』の「インターネットおよびXML アクセス用のステートメント」も参照してください。

スタティック変数の長さを決定することはできないため、PARSE ステートメントを使用するときはダイナミック変数を使用することをお勧めします。スタティック変数を使用すると、変数に書き込まれる値が切り捨てられる可能性があります。

Unicode サポートの詳細については、『Unicode およびコードページのサポート』ドキュメントの「PARSE XML」を参照してください。

### マークアップ

(ASCII ベースシステムで) XML ドキュメントの異なるデータタイプを表すためにパス文字列で使用されるマーキングを次に示します。

マーキング	XML データ	パス文字列内の位置
?	処理命令 (<?XML...?> を除く)	末尾
!	コメント	末尾
C	CDATA セクション	末尾
@	属性 (メインフレームでは §)	属性名の前
/	終了タグまたはパス内の親名セパレータ	末尾または親名の間
\$	解析したデータ - データ文字列	末尾

パス文字列でこの追加のマークアップを使用することにより、出力ドキュメントでXML ドキュメントの各種要素をより容易に識別できます。

## グローバル名前空間

グローバル名前空間を指定するには、接頭辞としてのコロン (:) および空の URI を使用してください。

## 関連するシステム変数

次の Natural システム変数は、発行された PARSE XML ステートメントごとに自動的に作成されます。

- \*PARSE-TYPE
- \*PARSE-LEVEL
- \*PARSE-ROW
- \*PARSE-COL
- \*PARSE-NAMESPACE-URI

\*PARSE-TYPE、\*PARSE-LEVEL、\*PARSE-ROW、\*PARSE-COL、および \*PARSE-NAMESPACE-URI の後の表記 (*r*) は、PARSE が発行されたステートメントのラベルまたはステートメント番号を示すために使用されます。( *r* ) 指定がない場合、対応するシステム変数は、アクティブな PARSE 処理ループで現在処理されている XML データのシステム変数を表します。

これらのシステム変数の詳細については、『システム変数』ドキュメントを参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A U B	可	不可
<i>operand2</i>	S	A U B	可	可
<i>operand3</i>	S	A U B	可	可
<i>operand4</i>	S	A U B	可	可
<i>operand5</i>	S A	A U B	可	可
<i>operand6</i>	S A	A U B	可	可

構文要素の説明：

<i>operand1</i>	<i>operand1</i> は対象の XML ドキュメントを表します。XML ドキュメントを解析中に変更することはできません。解析中に XML ドキュメントを変更（例えば、書き込み）しようとすると、エラーメッセージが表示されます。
<i>operand2</i>	<p><i>operand2</i> は XML ドキュメントのデータの PATH を表します。</p> <p>PATH には、識別された XML 部分の名前、すべての親の名前、および XML 部分のタイプが含まれます。</p> <p><b>注意:</b> PATH で指定された情報を使用して、ツリービューを容易に満たすことができます。</p> <p>「<a href="#">例1 - operand2 の使用</a>」も参照してください。</p>
<i>operand3</i>	<p><i>operand3</i> は XML ドキュメントのデータ要素の NAME を表します。</p> <p>NAME が値を持たない場合、それに関連する動的変数は *length()=0 (空白で満たされたスタティック変数) に設定されます。</p> <p>「<a href="#">例2 - operand3 の使用</a>」も参照してください。</p>
<i>operand4</i>	<p><i>operand4</i> は XML ドキュメントのデータ要素の内容 (VALUE) を表します。</p> <p>値がない場合、所定の動的変数は *length()=0, (空白で満たされたスタティック変数) に設定されます。</p> <p>「<a href="#">例3 - operand4 の使用</a>」も参照してください。</p>
<i>operand5</i> および	NAMESPACE URI または URI ( <i>operand5</i> ) および名前空間 PREFIX ( <i>operand6</i> ) はランタイム中にコピーされます。したがって、PARSE XML ループ内の名前空間マッピング配列の修正はパーサに影響しません。
<i>operand6</i>	<i>operand5</i> および <i>operand6</i> はオカレンス数の等しい 1 次元配列です。
<b>NORMALIZE NAMESPACE PREFIX</b>	<p>名前空間正規化は PARSE ステートメントの機能です。XML では、要素名に対応する名前空間を定義できます。</p> <pre>&lt;myns:myentity xmlns:myns="http://myuri" /&gt;</pre> <p>NAMESPACE 定義は次の部分で構成されます。</p> <ul style="list-style-type: none"> <li>■ 名前空間 PREFIX (この場合、myns)</li> <li>■ 名前空間を定義する URI (myuri)</li> </ul> <p>名前空間 PREFIX は要素名の一部です。つまり、PARSE ステートメントに対して、そして特に <i>operand2</i> に対して、生成された PATH 文字列は名前空間 PREFIX に依存します。Natural プログラム内のパスが特定のタグを示すために使用される場合、XML ドキュメントで正しい NAMESPACE (URI) を異なる PREFIX とともに使用すると失敗します。</p> <p>名前空間正規化では、すべての名前空間 PREFIX は、NAMESPACE 節で定義されているデフォルト値に設定できます。最初のエンタリは、URI が複数回指定される場合に使用されます。XML ドキュメントで複数の PREFIX が使用される場合、最初のものだけが出力に考慮されます。残りは無視されます。</p>

NAMESPACE 節には、名前空間 URI と接頭辞のペアが含まれます。次に例を示します。

```
uri(1) := 'http://namespaces.softwareag.com/natural/demo'  
pre(1) := 'nat:'
```

XML ドキュメント内に NAMESPACE が定義されている場合、パーサによってその名前空間 (URI) が正規化テーブルに存在するかどうかチェックされます。正規化テーブルの接頭辞は、XML ドキュメントに定義された名前空間の代わりに、PARSE ステートメントからのすべての出力データに対して使用されます。

以下の項目も参照してください。

- [例 4 - operand5 と operand6 の使用](#)
- [例 5 - 名前空間正規化での operand5 と operand6 の使用](#)

**PREFIX** に関する追加情報：

また、接頭辞定義には次の点が適用されます。

- 名前空間正規化配列の接頭辞定義は、置換される文字列であるため、常にコロン (:) で終了する必要があります。
- PREFIX または URI は、名前空間正規化配列には 1 回だけ存在できます。
- PREFIX または NAMESPACE URI に末尾の空白が含まれる場合 (例えば、スタティック変数の使用時)、外部パーサが呼び出される前に、末尾の空白は削除されます。
- 名前空間正規化での PREFIX 定義にコロン (:) のみが含まれる場合、NAMESPACE PREFIX は削除されます。

## 例

- [例 1 - operand2 の使用](#)
- [例 2 - operand3 の使用](#)
- [例 3 - operand4 の使用](#)
- [例 4 - operand5 と operand6 の使用](#)

■ 例 5 - 名前空間正規化でのoperand5 とoperand6 の使用

例 1 - operand2 の使用

次のような XML コードがあるとします。

```
myxml := '<?xml version="1.0" ?>'-
        '<employee personnel-id="30016315" >'-
        '<full-name>'-
        '<!--this is just a comment-->'-
        '<first-name>RICHARD</first-name>'-
        '<name>FORDHAM</name>'-
        '</full-name>'-
        '</employee>'
```

これは、次の Natural コードで処理されます。

```
PARSE XML myxml INTO PATH mypath
PRINT mypath
END-PARSE
```

次の出力を生成します。

```
employee
employee/@personnel-id
employee/full-name
employee/full-name/!
employee/full-name/first-name
employee/full-name/first-name/$
employee/full-name/first-name//
employee/full-name/name
employee/full-name/name/$
employee/full-name/name//
employee/full-name//
employee//
```



## 例 2 - operand3 の使用

次のような XML コードがあるとします。

```
myxml := '<?xml version="1.0" ?>'-
        '<employee personnel-id="30016315" >'-
        '<full-name>'-
        '<!--this is just a comment-->'-
        '<first-name>RICHARD</first-name>'-
        '<name>FORDHAM</name>'-
        '</full-name>'-
        '</employee>'
```

これは、次の Natural コードで処理されます。

```
PARSE XML myxml INTO PATH mypath NAME myname
  DISPLAY (AL=39) mypath myname
END-PARSE
```



**注意:** 次の出力を生成します。

MYPATH	MYNAME
employee	employee
employee/@personnel-id	personnel-id
employee/full-name	full-name
employee/full-name/!	
employee/full-name/first-name	first-name
employee/full-name/first-name/\$	
employee/full-name/first-name//	first-name
employee/full-name/name	name
employee/full-name/name/\$	
employee/full-name/name//	name
employee/full-name//	full-name
employee//	employee

### 例 3 - operand4 の使用

次のような XML コードがあるとします。

```
myxml := '<?xml version="1.0" ?>'-
        '<employee personnel-id="30016315" >'-
        '<full-name>'-
        '<!--this is just a comment-->'-
        '<first-name>RICHARD</first-name>'-
        '<name>FORDHAM</name>'-
        '</full-name>'-
        '</employee>'
```

これは、次の Natural コードで処理されます。

```
PARSE XML myxml INTO PATH mypath VALUE myvalue
  DISPLAY (AL=39) mypath myvalue
END-PARSE
```

次の出力を生成します。

MYPATH	MYVALUE
employee	
employee/@personnel-id	30016315
employee/full-name	
employee/full-name/!	this is just a comment
employee/full-name/first-name	
employee/full-name/first-name/\$	RICHARD
employee/full-name/first-name//	
employee/full-name/name	
employee/full-name/name/\$	FORDHAM
employee/full-name/name//	
employee/full-name//	
employee//	

## 例 4 - operand5 と operand6 の使用

次のような XML コードがあるとします。

```
myxml := '<?xml version="1.0" ?>'-
        '<nat:employee nat:personnel-id="30016315" '-
        '  xmlns:nat="http://namespaces.softwareag.com/natural/demo">'-
        '<nat:full-Name>'-
        '<nat:first-name>RICHARD</nat:first-name>'-
        '<nat:name>FORDHAM</nat:name>'-
        '</nat:full-Name>'-
        '</nat:employee>'
```

これは、次の Natural コードで処理されます。

```
PARSE XML myxml INTO PATH mypath
  PRINT mypath
END-PARSE
```

次の出力を生成します。

```
nat:employee
nat:employee/@nat:personnel-id
nat:employee/@xmlns:nat
nat:employee/nat:full-Name
nat:employee/nat:full-Name/nat:first-name
nat:employee/nat:full-Name/nat:first-name/$
nat:employee/nat:full-Name/nat:first-name//
nat:employee/nat:full-Name/nat:name
nat:employee/nat:full-Name/nat:name/$
nat:employee/nat:full-Name/nat:name//
nat:employee/nat:full-Name//
nat:employee//
```

## 例 5 - 名前空間正規化での operand5 と operand6 の使用

**NORMALIZE NAMESPACE** を使用すると、例 4 のような NAMESPACE PREFIX の異なる同じ XML ドキュメントでまったく同じ出力が生成されます。

XML コード :

```
myxml := '<?xml version="1.0" ?>' -
        '<natural:employee natural:personnel-id="30016315"' -
        ' xmlns:natural="http://namespaces.softwareag.com/natural/demo">' -
        '<natural:full-Name>' -
        '<natural:first-name>RICHARD</natural:first-name>' -
        '<natural:name>FORDHAM</natural:name>' -
        '</natural:full-Name>' -
        '</natural:employee>'
```

Natural コード :

```
uri(1) := 'http://namespaces.softwareag.com/natural/demo'
pre(1) := 'nat:'
*
PARSE XML myxml INTO PATH mypath NORMALIZE NAMESPACE uri(*) PREFIX pre(*)
PRINT mypath
END-PARSE
```

上記プログラムの出力 :

```
nat:employee
nat:employee/@nat:personnel-id
nat:employee/@xmlns:nat
nat:employee/nat:full-Name
nat:employee/nat:full-Name/nat:first-name
nat:employee/nat:full-Name/nat:first-name/$
nat:employee/nat:full-Name/nat:first-name//
nat:employee/nat:full-Name/nat:name
nat:employee/nat:full-Name/nat:name/$
nat:employee/nat:full-Name/nat:name//
nat:employee/nat:full-Name//
nat:employee//
```

# 92 PASSW

---

▪ 機能 .....	612
▪ 制限事項 .....	613
▪ 構文説明 .....	613
▪ 例 .....	614

PASSW=*operand1*

関連ステートメント：**ACCEPT/REJECT** | **AT BREAK** | **AT START OF DATA** | **AT END OF DATA** | **BACKOUT TRANSACTION** | **BEFORE BREAK PROCESSING** | **DELETE** | **END TRANSACTION** | **FIND** | **HISTOGRAM** | **GET** | **GET SAME** | **GET TRANSACTION** | **LIMIT** | **PERFORM BREAK PROCESSING** | **READ** | **RETRY** | **STORE** | **UPDATE**

関連機能グループ：「データベースへのアクセスと更新」

## 機能

PASSW ステートメントは、パスワード保護された Adabas または VSAM ファイルへのアクセスに必要なデフォルトのパスワードを指定するために使用します。



**注意:** このパスワードは、データベースアクセスステートメント **FIND**、**GET**、**HISTOGRAM**、**READ**、**STORE** の **PASSWORD** 節を使用して上書きできます。

### Natural Security の考慮事項

ライブラリのセキュリティプロファイルで、デフォルトの Adabas パスワード（『*Natural Security*』ドキュメントを参照）を指定できます。このパスワードは、個々パスワードも PASSW ステートメントも指定されていないすべてのデータベースアクセスステートメントに適用されます。これは、パスワードが指定されたセキュリティプロファイルを持つライブラリ内に適用されます。また、パスワードが指定されていないセキュリティプロファイルを持つ別のライブラリに後でログオンするときにも引き続き有効です

### パスワード表示の防止

パスワードを定数として指定する場合、PASSW ステートメントは必ずソースコード行の第1行目にコード化します。このとき、キーワード PASSW と等号の間に空白を含めないでください。こうすることで、プログラムのソースコードにパスワードは表示されません。

<b>TP モード:</b>	PASSW ステートメントを入力する前に端末コマンド %* を入力すれば、PASSW ステートメントは目に見えません。
----------------	---

バッチモード：	行エディタ
	EDT
	PASSW='password'
	END
	.E
	RUN
	で次のステートメントを指定することでパスワードを設定できます。パスワード値は出力には表示されません。

## 制限事項

このステートメントは、DL/I、DB2、および SQL/DS データベースには無効です。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A	可	不可

構文要素の説明：

<i>operand1</i>	<p>パスワード：</p> <p>パスワード (<i>operand1</i>) は、英数字定数または英数字変数の内容として指定できます。最大 8 文字から成り、特殊文字または埋め込み空白は使用できません。パスワードを定数として指定する場合は、一重引用符で囲む必要があります。</p> <p>PASSW ステートメントで指定したパスワードは、個々のパスワードが指定されていないすべてのデータベースアクセスステートメント (<b>FIND</b>、<b>GET</b>、<b>HISTOGRAM</b>、<b>READ</b>、<b>STORE</b>) に適用されます。一度指定したパスワードは、次の PASSW ステートメントの実行で別のパスワードが指定されるまで、または Natural セッションが終了するまで有効です。</p> <p>特定のデータベースアクセスステートメントで指定したパスワードは、そのステートメントにのみ適用され、他の後続のステートメントには適用されません。</p>
-----------------	--

## 例

パスワード表示を防止（[上記参照](#)）したプログラム例 PWDEX1

```
** Example 'PWDEX1': PASSW
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
END-DEFINE
*
PASSW=←——Password not visible
*
LIMIT 5
READ EMPLOY-VIEW
  DISPLAY NOTITLE PERSONNEL-ID NAME
END-READ
*
END
```

プログラム **PWDEX1** の出力：

PERSONNEL ID	NAME
50005800	ADAM
50005600	MORENO
50005500	BLOND
50005300	MAIZIERE
50004900	CAUDAL



# 93      PERFORM

---

▪ 機能 .....	616
▪ 構文説明 .....	617
▪ 例 .....	619

PERFORM <i>subroutine-name</i>	$\left[ \begin{array}{c} \textit{operand2} \left[ \left( \text{AD} = \left\{ \begin{array}{c} \text{M} \\ 0 \\ \text{A} \end{array} \right\} \right) \right] \\ \textit{nX} \end{array} \right]$	...

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CALL](#) | [CALL FILE](#) | [CALL LOOP](#) | [CALLNAT](#) | [DEFINE SUBROUTINE](#)  
| [ESCAPE](#) | [FETCH](#)

関連機能グループ：「[プログラムおよびルーチンの呼び出し](#)」

## 機能

PERFORM ステートメントは、Natural [サブルーチン](#)を呼び出すために使用します。

### PERFORM ステートメントのネスト構造

呼び出し先サブルーチン内で他のサブルーチンを呼び出すための PERFORM ステートメントを指定することもできます（ネストのレベル数は、要求されるメモリのサイズによって制限されます）。

サブルーチンは自分自身を呼び出すことができます（再帰サブルーチン）。再帰的に呼び出される外部サブルーチン内にデータベース処理が含まれている場合、Naturalはデータベース処理が論理的に分割されるようにします。

### ダイナミック変数を使用したパラメータ引き渡し

[CALLNAT](#) ステートメントを参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	動的定義
<i>operand2</i>	C S A G	A U N P I F B D T L C G O	可	可

構文要素の説明：

<i>subroutine-name</i>	<p>呼び出すサブルーチン：</p> <p>サブルーチン名（最大 32 文字）については、ユーザー定義変数と同じ命名規則が適用されます。</p> <p>サブルーチン名は、該当のサブルーチンが定義されているモジュールの名前に依存しません（同じ名前にすることは可能ですが、同じ名前にする必要はありません）。</p> <p>呼び出すサブルーチンは、DEFINE SUBROUTINE ステートメントで定義しておく必要があります。インラインサブルーチンまたは外部サブルーチンとして指定できます（<a href="#">DEFINE SUBROUTINE</a> ステートメントを参照）。</p> <p>1つのオブジェクト内で 50 個までの外部サブルーチンを参照できます。</p> <p>サブルーチンで使用可能なデータ</p> <ul style="list-style-type: none"> <li>■ インラインサブルーチン           <p>呼び出し側オブジェクトからインラインサブルーチンに明示パラメータを渡すことはできません。インラインサブルーチンは、同一オブジェクトモジュール内に定義されたローカルデータエリアや現在設定されているグローバルデータエリアにアクセスできます。</p> </li> <li>■ 外部サブルーチン           <p>外部サブルーチンは、現在設定されているグローバルデータエリアにアクセスできます。さらに、PERFORM ステートメントで呼び出し側オブジェクトから外部サブルーチンにパラメータを渡すことができるため（<i>operand2</i>を参照）、グローバルデータエリアのサイズを削減できます。</p> </li> </ul>
<i>operand2</i>	<p>外部サブルーチンへのパラメータの引き渡し：</p> <p>PERFORM ステートメントで外部サブルーチンを呼び出すとき、PERFORM ステートメントで呼び出し側オブジェクトから外部サブルーチンに 1つ以上のパラメータ（<i>operand2</i>）を渡すことができます。インラインサブルーチンでは、<i>operand2</i>は指定できません。</p> <p>パラメータを渡す場合、<a href="#">DEFINE DATA</a> ステートメントでパラメータリストの構造を定義する必要があります。</p>

	<p>デフォルトでは、パラメータは「参照によって」渡されます。つまり、データはアドレスパラメータを介して転送され、パラメータ値自体は移動されません。ただし、パラメータを「値によって」渡す、つまり実際のパラメータ値を渡すこともできます。そのためには、これらのフィールドをサブルーチンの <b>DEFINE DATA PARAMETER</b> ステートメントに BY VALUE または BY VALUE RESULT オプション付きで定義します。</p> <ul style="list-style-type: none"> <li>■ パラメータが「参照によって」渡される場合、次のことが適用されます。呼び出し側オブジェクトのパラメータの順序、フォーマット、および長さは、呼び出し先サブルーチンの <b>DEFINE DATA PARAMETER</b> 構造の順序、フォーマット、および長さと正確に一致している必要があります。変数の名前が呼び出し側オブジェクトとサブルーチンで異なる場合があります。</li> <li>■ パラメータが「値によって」渡される場合、次のことが適用されます。呼び出し側オブジェクトのパラメータの順序は、呼び出し先サブルーチンの <b>DEFINE DATA PARAMETER</b> 構造の順序と正確に一致している必要があります。呼び出し側オブジェクトの変数のフォーマットと長さは、サブルーチンの変数と異なっていてもかまいません。ただし、データ転送の互換性が必要です。変数の名前が呼び出し側オブジェクトとサブルーチンで異なる場合があります。サブルーチンで変更されたパラメータ値を呼び出し側オブジェクトに戻す場合、これらのフィールドを BY VALUE RESULT 付きで定義する必要があります。BY VALUE (RESULT なし) では、AD 指定 (下記も参照) に関係なく、変更したパラメータ値を呼び出し側オブジェクトに戻すことはできません。</li> </ul> <p><b>注意:</b> BY VALUE では、パラメータ値の内部コピーが作成されます。サブルーチンはこのコピーにアクセスし、それを変更できますが、その変更は呼び出し側オブジェクトの元のパラメータ値には影響しません。BY VALUE RESULT では、同様に内部コピーが作成されます。ただし、サブルーチンの終了後、元のパラメータ値はコピーの (変更された) 値で上書きされます。</p> <p>いずれのパラメータ引き渡し方法にも、次が適用されます。</p> <ul style="list-style-type: none"> <li>■ 呼び出し先サブルーチンのパラメータデータエリアでは、グループの再定義は REDEFINE ブロック内でのみ許可されます。</li> <li>■ 配列を渡す場合、サブルーチンのパラメータデータエリア内の次元数とオカレンス数は PERFORM パラメータリストと同じにする必要があります。</li> </ul> <p><b>注意:</b> インデックス付きグループの一部として定義された配列の複数オカレンスを PERFORM ステートメントで渡す場合、サブルーチンのパラメータデータエリア内の対応するフィールドを再定義しないでください。再定義すると、不正なアドレスが渡されます。</p>	
<p><b>AD=</b></p>	<p>属性の定義:</p> <p><i>operand2</i> が変数の場合は、次のいずれかの方法でマークすることができます。</p> <p>AD=O</p>	<p>変更不可。セッションパラメータ AD=O を参照してください。</p> <p><b>注意:</b> 内部的に、AD=O は BY VALUE と同じ方法で処理されます (<i>operand2</i> の注を参照)。</p>

	AD=M	変更可。セッションパラメータ AD=M を参照してください。  これはデフォルト設定です。
	AD=A	入力のみ。セッションパラメータ AD=A を参照してください。
	<i>operand2</i> が定数の場合は、AD を明示的に指定することはできません。定数には常に AD=0 が適用されます。	
<i>nX</i>	<p>省略されるパラメータの指定：</p> <p>表記 <i>nX</i> を使用して、次の <i>n</i> パラメータを省略するように指定できます。例えば、1X は次のパラメータを省略し、3X は次の3つのパラメータを省略します。これは、次の <i>n</i> パラメータに、外部サブルーチンに渡す値がないことを意味します。</p> <p>省略されるパラメータは、サブルーチンの <b>DEFINE DATA PARAMETER</b> ステートメントでキーワード <b>OPTIONAL</b> を使用して定義する必要があります。OPTIONAL は、値を呼び出し側オブジェクトからこのようなパラメータに渡すこともできるということを意味します。</p>	

## 例

- 例 1- インラインサブルーチンとしての PERFORM
- 例 2- 外部サブルーチンとしての PERFORM

### 例 1- インラインサブルーチンとしての PERFORM

```

** Example 'PEREX1': PERFORM (as inline subroutine)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 ADDRESS-LINE (A20/2)
  2 PHONE
*
1 #ARRAY (A75/1:4)
1 REDEFINE #ARRAY
  2 #ALINE (A25/1:4,1:3)
1 #X (N2) INIT <1>
1 #Y (N2) INIT <1>
END-DEFINE
*
LIMIT 5
FIND EMPLOY-VIEW WITH CITY = 'BALTIMORE'
  MOVE NAME TO #ALINE (#X,#Y)
  MOVE ADDRESS-LINE(1) TO #ALINE (#X+1,#Y)
  MOVE ADDRESS-LINE(2) TO #ALINE (#X+2,#Y)

```

## PERFORM

```
MOVE PHONE          TO #ALINE (#X+3,#Y)
IF #Y = 3
  RESET INITIAL #Y
  /*
  PERFORM PRINT
  /*
ELSE
  ADD 1 TO #Y
END-IF
AT END OF DATA
  /*
  PERFORM PRINT
  /*
END-ENDDATA
END-FIND
*
DEFINE SUBROUTINE PRINT
  WRITE NOTITLE (AD=OI) #ARRAY(*)
  RESET #ARRAY(*)
  SKIP 1
END-SUBROUTINE
*
END
```

プログラム PEREX1 の出力：

```
JENSON                LAWLER                FORREST
2120 HASSELL          4588 CANDLEBERRY AVE 37 TENNYSON DRIVE
#206                  BALTIMORE              BALTIMORE
998-5038              629-0403              881-3609

ALEXANDER             NEEDHAM
409 SENECA DRIVE      12609 BUILDERS LANE
BALTIMORE             BALTIMORE
345-3690              641-9789
```

## 例 2 - 外部サブルーチンとしての PERFORM

PERFORM ステートメントを含むプログラム：

```
** Example 'PEREX2': PERFORM (as external subroutine)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 ADDRESS-LINE (A20/2)
  2 PHONE
*
1 #ALINE (A25/1:4,1:3)
1 #X (N2)          INIT <1>
```

```

1 #Y      (N2)          INIT <1>
END-DEFINE
*
LIMIT 5
*
FIND EMPLOY-VIEW WITH CITY = 'BALTIMORE'
  MOVE NAME           TO #ALINE (#X,#Y)
  MOVE ADDRESS-LINE(1) TO #ALINE (#X+1,#Y)
  MOVE ADDRESS-LINE(2) TO #ALINE (#X+2,#Y)
  MOVE PHONE          TO #ALINE (#X+3,#Y)
  IF #Y = 3
    RESET INITIAL #Y
    /*
    PERFORM PEREX2E #ALINE(*,*)
    /*
  ELSE
    ADD 1 TO #Y
  END-IF
  AT END OF DATA
  /*
  PERFORM PEREX2E #ALINE(*,*)
  /*
  END-ENDDATA
END-FIND
*
END

```

プログラム PEREX2 から呼び出されるパラメータを含む外部サブルーチン PEREX3 :

```

** Example 'PEREX3': SUBROUTINE (external subroutine with parameters)
*****
DEFINE DATA
PARAMETER
1 #ALINE (A25/1:4,1:3)
END-DEFINE
*
DEFINE SUBROUTINE PEREX2E
  WRITE NOTITLE (AD=OI) #ALINE(*,*)
  RESET #ALINE(*,*)
  SKIP 1
END-SUBROUTINE
*
END

```

プログラム PEREX2 の出力 :

## PERFORM

---

JENSON  
2120 HASSELL  
#206  
998-5038

LAWLER  
4588 CANDLEBERRY AVE  
BALTIMORE  
629-0403

FORREST  
37 TENNYSON DRIVE  
BALTIMORE  
881-3609

ALEXANDER  
409 SENECA DRIVE  
BALTIMORE  
345-3690

NEEDHAM  
12609 BUILDERS LANE  
BALTIMORE  
641-9789



# 94 PERFORM BREAK PROCESSING

---

■ 機能 .....	624
■ 構文説明 .....	624
■ 例 .....	625

```
PERFORM BREAK [PROCESSING] [(r)]
AT BREAK statement ...
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION DATA](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [READ](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

PERFORM BREAK PROCESSING ステートメントは、自動ブレイク処理が設定されない [FOR](#)、[REPEAT](#)、[CALL LOOP](#)、および [CALL FILE](#) ステートメントで生成されるループ内にブレイク処理を設定するために使用します。または、ユーザー開始のブレイク処理が必要な場合に使用します。レコードが読み取られた直後に実行される自動ブレイク処理とは異なり、PERFORM BREAK PROCESSING ステートメントは、プログラムの通常の流れの中で出現したときに初めて実行されます。

このステートメントでは、（コントロールフィールドの値に基づいて）ブレイク処理の条件をチェックし、Natural システム関数を評価します。このチェックおよびシステム関数の評価は、実行対象のステートメントが出現するたびに行われます。このステートメントは、[IF](#) ステートメントで指定した条件に応じて実行することもできます。

## 構文説明

(r)	<p>ステートメント参照表記：</p> <p>デフォルトでは、プログラム、サブプログラム、またはサブルーチンの実行終了時に最後の PERFORM BREAK 条件が真となります。</p> <p>表記 (r) を使用して、PERFORM BREAK の最後の処理を特定のループに関連付けることができます。この場合、PERFORM BREAK はそのループの終了処理の中（最後の自動 BREAK 処理後かつ <a href="#">AT END OF DATA</a> ステートメントの実行前）で実行されます。</p>
<a href="#">AT BREAK</a> <i>statement...</i>	<a href="#">AT BREAK</a> ステートメントの構文を参照してください。

## 例

```

** Example 'PBPEX1S': PERFORM BREAK PROCESSING (structured mode)
*****
DEFINE DATA LOCAL
1 #INDEX (N2)
1 #LINE (N2) INIT <1>
END-DEFINE
*
FOR #INDEX 1 TO 18
  PERFORM BREAK PROCESSING
  /*
  AT BREAK OF #INDEX /1/
    WRITE NOTITLE / 'PLEASE COMPLETE LINES 1-9 ABOVE' /
    RESET INITIAL #LINE
  END-BREAK
  /*
  WRITE NOTITLE '_' (64) '=' #LINE
  ADD 1 TO #LINE
END-FOR
*
END

```

プログラム **PBPEX1S** の出力：

```

#LINE: 1
#LINE: 2
#LINE: 3
#LINE: 4
#LINE: 5
#LINE: 6
#LINE: 7
#LINE: 8
#LINE: 9

PLEASE COMPLETE LINES 1-9 ABOVE

#LINE: 1
#LINE: 2
#LINE: 3
#LINE: 4
#LINE: 5
#LINE: 6
#LINE: 7
#LINE: 8
#LINE: 9

PLEASE COMPLETE LINES 1-9 ABOVE


```

レポートモードの例はライブラリ SYSEXRM のプログラム [PBPEX1R](#) を参照してください。

# 95 PRINT

---

■ 機能 .....	628
■ 構文説明 .....	629
■ 例 .....	634



構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT END OF PAGE](#) | [AT TOP OF PAGE](#) | [CLOSE PRINTER](#) | [DEFINE PRINTER](#) | [DISPLAY](#) | [EJECT](#) | [FORMAT](#) | [NEWPAGE](#) | [SKIP](#) | [SUSPEND IDENTICAL SUPPRESS](#) | [WRITE](#) | [WRITE TITLE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

PRINT ステートメントは、出力をフリーフォーマットで生成するために使用します。

PRINT ステートメントは、次の点で [WRITE](#) ステートメントと異なります。

- 各オペランドの出力は、オペランドの長さではなく値の内容に合わせて書き込まれます。数値の先頭のゼロおよび英数字の末尾の空白は省略されます。セッションパラメータ AD では、数値を左揃えで出力するか右揃えで出力するかを定義します。AD=L の場合、数値の末尾の空白は省略されます。AD=R の場合、数値の先頭の空白は出力されます。
- 出力結果が現在の行サイズ（LS パラメータ）を超過した場合、出力が次の行に続きます。このとき、英数字定数、または編集マスク指定のない英数字変数の内容は、現在の行の右端の空白または英字でも数字でもない文字で分割されます。分割された値の初めの部分は現在の行に出力され、後ろの部分は次の行に出力されます。2番目の部分の先頭の空白は削除されず。この結果、空の行は非表示になります。

他のすべてのオペランドについては、値全体が次の行に出力されます。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A G N A U N P I F B D T L G O		可	不可

構文要素の説明：

( <i>rep</i> )	<p>レポート指定：</p> <p>表記 (<i>rep</i>) を使用して、PRINT ステートメントを適用するレポートの ID を指定できます。</p> <p>範囲 0～31 の値、または <b>DEFINE PRINTER</b> ステートメントを使用して割り当てた論理名を指定できます。</p> <p>(<i>rep</i>) 指定がない場合、PRINT ステートメントは最初のレポート（レポート 0）に適用されます。</p> <p>このプリンタファイルが PC として Natural に定義されている場合、レポートは PC にダウンロードされます。<a href="#">例 2</a> を参照してください。</p> <p>Natural で作成される出力レポートの形式を制御する方法については、『プログラミングガイド』の「データ出力制御」を参照してください。</p>
NOTITLE	<p>デフォルトページタイトルの省略：</p> <p>Natural は、PRINT ステートメントによって出力される各ページに 1 行のタイトル行を生成します。このタイトルには、ページ番号と日時が含まれます。時刻は、セッション開始時（TP モード）またはジョブ開始時（バッチモード）に設定されます。このデフォルトのタイトル行は <b>WRITE TITLE</b> ステートメントを使用して上書きできます。または、PRINT ステートメントに NOTITLE 節を指定することで省略することもできます。例：</p> <ul style="list-style-type: none"> <li>■ デフォルトタイトルが生成されます。</li> </ul> <pre>PRINT NAME</pre>

	<p>■ ユーザータイトルが生成されます。</p> <pre>PRINT NAME WRITE TITLE 'user-title'</pre> <p>■ タイトルは生成されません。</p> <pre>PRINT NOTITLE NAME</pre> <p>NOTITLE オプションを使用すると、同じオブジェクト内でデータを同じレポートに書き込むすべての <b>DISPLAY</b>、<b>PRINT</b>、および <b>WRITE</b> ステートメントに適用されます。</p>
<p><b>NOHDR</b></p>	<p>列ヘッダーの省略：</p> <p>PRINT ステートメント自体では、列ヘッダーは生成されません。ただし、PRINT ステートメントを <b>DISPLAY</b> ステートメントと組み合わせて使用する場合、PRINT ステートメントの <b>NOHDR</b> オプションを使用して、<b>DISPLAY</b> ステートメントによって生成される列ヘッダーを省略できます。NOHDR オプションは、PRINT ステートメントの実行によって新しいページが出力される場合にのみ有効です。</p> <p>NOHDR オプションを使用しないと、この新しいページに <b>DISPLAY</b> ステートメントの列ヘッダー（存在する場合）が出力されます。NOHDR を使用すると、出力されません。</p>
<p><i>statement-parameters</i></p>	<p>ステートメントレベルでのパラメータ定義：</p> <p>1つまたは複数のパラメータをカッコで囲んで、ステートメントレベルで、つまり PRINT ステートメントまたは出力要素の直後に指定できます。</p> <p>この方法で指定した各パラメータは、以前に <b>GLOBALS</b> コマンド、<b>SET GLOBALS</b> ステートメント（レポートモードでのみ）、または <b>FORMAT</b> ステートメントで指定したパラメータを上書きします。複数のパラメータを指定する場合は、各パラメータ間を1つ以上の空白で区切る必要があります。1つのパラメータエントリを2行に分けて指定することはできません。</p> <p>ここで適用されるパラメータ設定は、変数フィールドにのみ関連し、テキスト定数には影響しません。テキスト定数にフィールド属性を設定する場合は、この要素に属性を明示的に設定する必要があります（「<a href="#">要素（フィールド）レベルでのパラメータ定義</a>」を参照）。</p> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <a href="#">パラメータのリスト</a></li> <li>■ <a href="#">ステートメントおよび要素（フィールド）レベルでのパラメータ使用例</a></li> </ul>
<p><i>nX, nT, l</i></p>	<p>下記の「<a href="#">フィールドの位置指定、テキスト／属性の割り当て</a>」を参照してください。</p>



## パラメータのリスト

PRINT ステートメントで指定可能なパラメータ		指定 (S=ステートメントレベル、E=要素レベル)
AD	属性定義	SE
AL	出力の英数字長	SE
CD	カラー定義	SE
CV	制御変数	SE
DF	日付フォーマット	SE
DL	出力の表示長	SE
DY	ダイナミック属性	SE
EM	編集マスク	SE
FL	浮動小数点仮数長	SE
MC	マルチプルバリューフィールド数	S
MP	レポートの最大ページ数	S
NL	出力の数値長	SE
PC	ピリオディックグループ数	S
PM	出力モード	SE
SG	符号の位置	SE
ZP	ゼロ出力	SE

各セッションパラメータの詳細については、『パラメータリファレンス』を参照してください。

### ステートメントおよび要素（フィールド）レベルでのパラメータ使用例

#### フィールドの位置指定、テキスト／属性の割り当て

```

{ [ [ nX ] | [ nT ] | [ / ] ... { 'text'[(attributes)] | 'c'(n) [(attributes)] | ['='] operand1 [(parameters)] } } ...

```

## フィールドの位置指定表記

<b><i>nX</i></b>	<p>列の間隔： この表記により、列の間に <i>n</i> 個のスペースが挿入されます。<i>n</i> をゼロにすることはできません。</p> <pre>PRINT NAME 5X SALARY</pre>
<b><i>nT</i></b>	<p>タブ設定： <i>nT</i> 表記により、位置 <i>n</i> を出力するように位置指定（タブ設定）されます。後方への位置指定は行送りになります。</p> <p>次の例では、NAME は位置 25 から、SALARY は位置 50 から出力されます。</p> <pre>PRINT 25T NAME 50T SALARY</pre>
<b><i>/</i></b>	<p>行送り - スラッシュ表記： フィールドまたはテキスト要素の間に配置した場合は、スラッシュ (<i>/</i>) により次の出力行の先頭に位置指定されます。</p> <pre>PRINT NAME / SALARY</pre>

## テキスト／属性割り当て

<b>'text'</b>	<p>テキスト割り当て： 一重引用符で囲まれた文字列が表示されます。</p> <pre>PRINT 'EMPLOYEE' NAME 'MARITAL/STATUS' MAR-STAT</pre>
<b>'c' (<i>n</i>)</b>	<p>文字の繰り返し： フィールド値の直前に、一重引用符で囲まれた文字が <i>n</i> 回表示されます。</p> <pre>PRINT '*' (5) '=' NAME</pre>

'='	<p>フィールドヘッダーの後に位置するフィールド内容：</p> <p>フィールドの前に配置した場合は、等号 '=' により、<b>DEFINE DATA</b> ステートメントまたは <b>DDM</b> の定義に従って、フィールドヘッダーはフィールド内容の前に表示されます。</p> <pre>PRINT '=' NAME</pre>
<i>operand1</i>	<p>出力するフィールド：</p> <p><i>operand1</i> として、出力するフィールドを指定します。</p>
<i>parameters</i>	<p>要素（フィールド）レベルでのパラメータ定義：</p> <p>1つまたは複数のパラメータ（上記の表を参照）をカッコで囲んで、<i>operand1</i> の直後に指定できます。</p> <p>この方法で指定した各パラメータは、以前に <b>ステートメントレベル</b>、<b>GLOBALS</b> コマンド、<b>SET GLOBALS</b> ステートメント（レポートモードでのみ）、または <b>FORMAT</b> ステートメントで指定したパラメータを上書きします。</p> <p>複数のパラメータを指定する場合は、各エントリ間に1つ以上の空白を配置する必要があります。エントリを2行のステートメント行に分割することはできません。</p> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <a href="#">ステートメントのパラメータ</a></li> <li>■ <a href="#">ステートメントおよび要素（フィールド）レベルでのパラメータ使用例</a></li> </ul>

## 出力属性

*attributes* は、テキスト表示に使用される出力属性を示します。可能な属性：

$\left\{ \begin{array}{l} \left\{ \begin{array}{l} AD=AD-value \dots \\ BX=BX-value \dots \\ CD=CD-value \dots \\ PM=PM-value \dots \end{array} \right\} \dots \end{array} \right\}$	$\left\{ \begin{array}{l} \left\{ \begin{array}{l} AD-value \dots \\ CD-value \dots \end{array} \right\} \dots \end{array} \right\}$
--	--

指定可能なセッションパラメータ値については、『パラメータリファレンス』ドキュメントの該当するセクションを参照してください。

- 「AD - 属性定義」の「フィールド表現」
- CD - カラー定義
- BX - 外枠の定義
- PM - 出力モード



**注意:** コンパイラは、実際には1つの出力フィールドに複数の属性値を受け入れます。例えば、「AD=BDI」と指定できます。ただし、この場合は最後の値のみが適用されます。示した例では、値 I のみが有効になり、出力フィールドは強調表示されます。

## 例

- 例 1 - PRINT ステートメント
- 例 2 - レポートが PC にダウンロードされる PRINT ステートメント

### 例 1 - PRINT ステートメント

```

** Example 'PRTEX1': PRINT
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 CITY
  2 JOB-TITLE
  2 ADDRESS-LINE (2)
END-DEFINE
*
LIMIT 1
READ EMPLOY-VIEW BY CITY
/*
WRITE NOTITLE 'EXAMPLE 1:'
           // 'RESULT OF WRITE STATEMENT:'
WRITE      /  NAME  ',' FIRST-NAME ':' JOB-TITLE '*' (30)
WRITE      /  'RESULT OF PRINT STATEMENT:'
PRINT      /  NAME  ',' FIRST-NAME ':' JOB-TITLE '*' (30)
/*
WRITE      // 'EXAMPLE 2:'
           // 'RESULT OF WRITE STATEMENT:'
WRITE      /  NAME 60X ADDRESS-LINE (1:2)
WRITE      /  'RESULT OF PRINT STATEMENT:'
PRINT      /  NAME 60X ADDRESS-LINE (1:2)
/*
END-READ
END

```

プログラム **PRTXEX1** の出力：

EXAMPLE 1:

RESULT OF WRITE STATEMENT:

```
SENKO                , WILLIE                : PROGRAMMER
*****
```

RESULT OF PRINT STATEMENT:

```
SENKO , WILLIE : PROGRAMMER *****
```

EXAMPLE 2:

RESULT OF WRITE STATEMENT:

```
SENKO
2200 COLUMBIA PIKE    #914
```

RESULT OF PRINT STATEMENT:

```
SENKO                2200 COLUMBIA
PIKE #914
```

## 例 2 - レポートが PC にダウンロードされる PRINT ステートメント

```
** Example 'PCPIEX1': PRINT to PC
**
** NOTE: Example requires that Natural Connection is installed.
*****
DEFINE DATA LOCAL
01 PERS VIEW OF EMPLOYEES
   02 PERSONNEL-ID
   02 NAME
   02 CITY
END-DEFINE
*
FIND PERS WITH CITY = 'NEW YORK'           /* Data selection
   PRINT (7) 5T CITY 20T NAME 40T PERSONNEL-ID /* (7) designates
                                                    /* the output file
                                                    /* (here the PC).
END-FIND
END
```



# 96 PROCESS

---

■ 機能 .....	638
■ 制限事項 .....	638
■ 構文説明 .....	638

```
PROCESS view-name USING operand1=operand2 GIVING operand3...
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

PROCESS ステートメントは、Entire System Server と組み合わせて使用します。Entire System Server は、ファイルの読み書き、VTOC およびカタログ管理、JES キューなどの各種オペレーティングシステム機能の使用を可能にします。

PROCESS ステートメントと個々の節の詳細については、『*Entire System Server User's Guide*』の「*Getting Started*」セクションを参照してください。

## 制限事項

このステートメントは Entire System Server 以外では使用できません。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A N P B	可	不可
<i>operand2</i>	C S	A U N P B	可	不可
<i>operand3</i>	S	A N P B	可	不可

構文要素の説明：

<i>view-name</i>	Entire System Server で使用されるビューの名前です。
<b>USING</b>	<p>USING 節は、Entire System Server プロセッサにパラメータを渡すために使用します。そのためには、Entire System Server に定義されたビューで値 (<i>operand2</i>) をフィールド (<i>operand1</i>) に割り当てます。ビューについては、Entire System Server ドキュメントを参照してください。</p> <p><b>注意:</b> <i>operand1=operand2</i> の複数指定は、INPUT 区切り文字 (セッションパラメータ ID で指定) またはコンマで区切る必要があります。セッションパラメータ DC でコンマを小数点文字として定義している場合は、この目的でコンマを使用することはできません。</p>



---

<b>GIVING</b>	GIVING 節は、値が Entire System Server プロセッサによって返されるフィールド ( <i>operand3</i> ) を指定するために使用します。各フィールドは、Entire System Server で使用されるビューに定義する必要があります。
---------------	--



# 97 PROCESS COMMAND

---

▪ 機能 .....	643
▪ 構文説明 .....	643
▪ DDM : COMMAND .....	654
▪ 例 .....	655

ストラクチャードモード構文

PROCESS	COMMAND	ACTION			
}	{	CLOSE			
		CHECK			
		}	EXEC	USING	PROCESSOR-NAME= <i>operand1</i>
					COMMAND-LINE
					( <i>index[:index]</i> )= <i>operand2</i>
		}	TEXT		
			HELP		
			GET USING	PROCESSOR-NAME= <i>operand1</i>	
				GETSET-FIELD-NAME= <i>operand3</i>	
	SET USING	PROCESSOR-NAME= <i>operand1</i>			
		GETSET-FIELD-NAME= <i>operand3</i>			
		GETSET-FIELD-VALUE= <i>operand4</i>			

レポートイングモード構文

PROCESS	COMMAND	ACTION			
}	{	CLOSE	[GIVING NATURAL - ERROR]		
		CHECK			
		}	EXEC	USING	PROCESSOR-NAME= <i>operand1</i>
					COMMAND-LINE ( <i>index[:</i>
					<i>index</i> )= <i>operand2</i>
		}	TEXT		GIVINGRESULT-FIELD ( <i>index[:</i>
					<i>index</i> )]
			HELP		RETURN-CODE [NATURAL - ERROR]
			GET USING	PROCESSOR-NAME= <i>operand1</i>	
				GETSET-FIELD-NAME= <i>operand3</i>	
		GETSET-FIELD-VALUE= <i>operand4</i> [NATURAL - ERROR]			
	SET USING	PROCESSOR-NAME= <i>operand1</i>			
		GETSET-FIELD-NAME= <i>operand3</i>			
		GETSET-FIELD-VALUE= <i>operand4</i> [GIVING NATURAL - ERROR]			

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[プログラムおよびルーチンの呼び出し](#)」

## 機能

Natural ユーティリティ SYSNCP で作成したコマンドプロセッサは、PROCESS COMMAND ステートメントを使用して Natural プログラムから呼び出すことができます。

Natural コマンドプロセッサの作成方法の詳細については、『SYSNCP ユーティリティ』ドキュメントを参照してください。



**注意:** PROCESS COMMAND ステートメント内の単語「COMMAND」は実際のビュー名です。使用されるビューの名前は必ずしも「COMMAND」である必要はありませんが、**同じ名前の DDM** が存在するため、「COMMAND」を使用することをお勧めします。DEFINE DATA ステートメント（例えば、COMMAND VIEW OF COMMAND）内でこの DDM を参照する必要があります。

### セキュリティに関する考慮事項

Natural Security では、コマンドプロセッサで定義された特定のキーワードや機能の使用を制限できます。特定のユーザーまたはユーザーグループに対して、キーワードや機能の使用を許可／禁止できます。詳細については、『Natural Security』ドキュメントを参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A	不可	不可
<i>operand2</i>	C S A G	A N	不可	不可
<i>operand3</i>	C S	A N	不可	不可
<i>operand4</i>	C S	A N P I	不可	不可

構文要素の説明：

<b>CLOSE</b>	<p>CLOSE では、コマンドプロセッサの使用を終了し、コマンドプロセッサバッファを解放します。</p> <p>セッション中にコマンドプロセッサを使用し、CLOSE で解放しないと、ユーザーのスレッド中に NCPWORK という名前のバッファが存在します。このバッファのパラメータはシステムコマンド BUS で評価できます。コマンドプロセッサのランタイム部分でこのバッファが必要になります。このバッファは PROCESS COMMAND ACTION CLOSE ステートメントで解放できます。</p> <p>このステートメントに続けて PROCESS COMMAND ステートメントを指定すると、コマンドプロセッサバッファが再びオープンされます。</p>
--------------	---

	<p>「例1 - PROCESS COMMAND ACTION CLOSE」も参照してください。</p>
<b>CHECK</b>	<p>CHECK は、コマンドが PROCESS COMMAND EXEC ステートメントで実行可能かどうかを判断するための手段として使用します。指定プロセッサ名に対して、ランタイムチェックが2段階で行われます。</p> <ul style="list-style-type: none"> <li>■ 現在のライブラリまたは STEPLIB の1つにプロセッサが存在するかどうかをチェックします。</li> <li>■ コマンド行 COMMAND-LINE (1) の内容が受け入れ可能かどうかを判断するために分析します。</li> </ul> <p>または、ランタイムアクション定義 R、M、および 1~9 は RESULT-FIELD (1:9) に書き込まれます。</p> <p>NATURAL-ERROR フィールドをビューまたは GIVING 節で指定した場合、エラーコードが返されます。このフィールドが利用できず、コマンド分析が失敗すると、Natural システムエラーが発生します。</p> <p><b>注意:</b> CHECK オプションの機能は EXEC オプションの一部としても実行されるため、各 EXEC の前に CHECK を使用する必要はありません。</p>
<b>EXEC</b>	<p>EXEC の動作は CHECK とまったく同じですが、それに加え、ランタイムアクションエディタで指定されたランタイムアクションが実行されます。</p> <p>COMMAND-LINE (1) のみが必要です。RESULT-FIELD のオカレンスは9個まで使用できます（ただし、最適なパフォーマンスを得るためには、実際に必要なオカレンスのみを使用してください）。</p> <p><b>注意:</b> EXEC は、現在アクティブなプログラムから抜けるために使用できる唯一のオプションです。ランタイムアクション定義に FETCH または STOP ステートメントが含まれる場合が該当します。</p> <p>「例2 - PROCESS COMMAND ACTION EXEC」も参照してください。</p>
<b>HELP</b>	<p>HELP は、有効なすべてのキーワード、同義語、およびオンラインヘルプウィンドウの作成などの機能のリストを返します。このリストは RESULT-FIELD のフィールド（複数可）に含まれます。返されるヘルプのタイプはコマンド行の内容に依存します。</p> <ul style="list-style-type: none"> <li>■ COMMAND-LINE (1) には検索条件を含める必要があります。</li> <li>■ COMMAND-LINE (2) (指定する場合) には開始値または検索値を含める必要があります。</li> <li>■ COMMAND-LINE (3) (指定する場合) には開始値を含める必要があります。</li> </ul> <p>詳細については、以下のセクションを参照してください。</p> <ul style="list-style-type: none"> <li>■ キーワードの <b>HELP</b></li> <li>■ 同義語の <b>HELP</b></li> <li>■ グローバル機能の <b>HELP</b></li> <li>■ ローカル機能の <b>HELP</b></li> <li>■ <b>IKN</b> の <b>HELP</b></li> <li>■ <b>IFN</b> の <b>HELP</b></li> </ul>

	<p><b>注意:</b> 最適なパフォーマンスを得るためには、フィールド <b>RESULT-FIELD</b> のオカレンス数は、画面に表示する行数を超過しないようにする必要があります。最低1オカレンスを使用する必要があります。</p>
<b>TEXT</b>	<p>TEXT オプションは、プロセッサの全般情報およびキーワードや機能に関連するテキストを提供するために使用します。このテキストは、コマンドプロセッサ定義時に SYSNCP ユーティリティのキーワードエディタやアクションエディタに入力されたものと同じです。</p> <p>詳細については、以下のセクションを参照してください。</p> <ul style="list-style-type: none"> <li>■ 全般的な情報の <b>TEXT</b></li> <li>■ キーワード情報の <b>TEXT</b></li> <li>■ 機能情報の <b>TEXT</b></li> </ul> <p><b>注意:</b> キーワードや機能に関するテキストにアクセスするには、SYSNCP ユーティリティの [Processor Header Maintenance 3] 画面の Catalog user texts フィールドに Y が指定されている必要があります（「その他のオプション-ヘッダー3」セクションを参照）。</p>

## キーワードの HELP

このオプションは、アルファベット順にソートされたキーワードまたは同義語を IKN 付きで返します。

コマンド行	内容
1	インジケータ K で始める必要があります。
	次のキーワードタイプが返されます。
*	すべてのタイプのキーワード
1	タイプ 1 のキーワード
2	タイプ 2 のキーワード
3	タイプ 3 のキーワード
P	タイプ P (パラメータ) のキーワード
	<b>オプション:</b>
I	キーワードと IKN を返します。
T	キーワードの一部を大文字で示します (可能な省略形を示すため)。
S	同義語とキーワードを返します。
X	指定したキーワードの同義語のみを返します。
A	内部キーワードも返されます。
+	検索には開始値は含まれません。

コマンド行	内容
2	<p>キーワード検索の開始値（オプション）。</p> <p>デフォルトでは、検索は開始値から始まります。ただし、プラス (+) オプションを指定すると、開始値自体は検索に含まれず、次に高い値から検索されます。</p>

フィールド **RESULT-FIELD (1:n)** は、指定したリストを返します。

例：

Command Line 1: K\*X Returns all synonyms of all keyword types.

Command Line 1: K123S Returns all keywords of type 1, 2 and 3 including synonyms.

### 同義語の HELP

指定された IKN に対して、このオプションは元のキーワードとすべての同義語を返します。

コマンド行	内容	
1	インジケータ S で始める必要があります。	
	<p>オプション：</p> <table border="1"> <tr> <td>T</td> <td>キーワードの一部を大文字で示します（可能な省略形を示すため）。</td> </tr> </table>	T
T	キーワードの一部を大文字で示します（可能な省略形を示すため）。	
2	フォーマット N4 のキーワードの内部キーワード番号 (IKN) 。	

フィールド **RESULT-FIELD (1)** は、元のキーワードを返します。フィールド **RESULT-FIELD (2:n)** は、このキーワードに関連する同義語を返します。

例：

入力：	出力：
Command Line 1: S	Result-Field 1: Edit
Command Line 2: 1003	Result-Field 2: Maintain
	Result-Field 3: Modify



## グローバル機能の HELP

このオプションは、すべてのグローバル機能のリストを返します。

コマンド行	内容
1	インジケータ G で始める必要があります。
	オプション：
I	内部機能番号 (IFN : Internal Function Number) も返されます。
T	キーワードの一部を大文字で示します (可能な省略形を示すため)。
S	RESULT-FIELD に返されたキーワードは列に整列されます。
A	内部キーワードも返されます。
1	タイプ 1 の特定のキーワードを含む機能のみが返されます。
2	タイプ 2 の特定のキーワードを含む機能のみが返されます。
3	タイプ 3 の特定のキーワードを含む機能のみが返されます。
+	検索には開始値は含まれません。
2	グローバル機能検索の開始値。キーワードは 123 の順序で指定する必要があります。 デフォルトでは、検索は開始値から始まります。ただし、プラス (+) オプションを指定すると、開始値自体は検索に含まれず、次に高い値から検索されます。
3	空白にする必要があります。
4	特定のキーワードを含むグローバル機能のみを検索するには、ここにそのキーワードを指定します。 キーワードを指定する場合、オプション (上記参照) としてキーワードタイプ (1、2、または 3) を指定する必要があります。

フィールド RESULT-FIELD (1:n) は、指定したリストを返します。

例：

入力：	出力：
Command Line 1: G	Result-Field 1: ADD CUSTOMER
Command Line 2: ADD	Result-Field 2: ADD FILE
	Result-Field 3: ADD USER

### ローカル機能の HELP

このオプションは、指定したロケーションのすべてのローカル機能のリストを返します。

コマンド行	内容																		
1	<p>インジケータ L で始める必要があります。</p> <p>オプション：</p> <table border="1"> <tr> <td>I</td> <td>内部機能番号 (IFN: Internal Function Number) も返されません。</td> </tr> <tr> <td>T</td> <td>キーワードの一部を大文字で示します (可能な省略形を示すため)。</td> </tr> <tr> <td>S</td> <td>RESULT-FIELD に返されたキーワードは列に整列されます。</td> </tr> <tr> <td>A</td> <td>内部キーワードも返されます。</td> </tr> <tr> <td>1</td> <td>タイプ 1 の特定のキーワードを含む機能のみが返されます。</td> </tr> <tr> <td>2</td> <td>タイプ 2 の特定のキーワードを含む機能のみが返されます。</td> </tr> <tr> <td>3</td> <td>タイプ 3 の特定のキーワードを含む機能のみが返されます。</td> </tr> <tr> <td>C</td> <td>現在のロケーションに定義されている機能のみが返されます (コマンド行 3 は無視されます)。</td> </tr> <tr> <td>F</td> <td>ローカル機能の「再帰的な」リストを呼び出します。つまり、現在のロケーション/指定したロケーションに関連するすべてのローカルコマンドが返されます。</td> </tr> </table>	I	内部機能番号 (IFN: Internal Function Number) も返されません。	T	キーワードの一部を大文字で示します (可能な省略形を示すため)。	S	RESULT-FIELD に返されたキーワードは列に整列されます。	A	内部キーワードも返されます。	1	タイプ 1 の特定のキーワードを含む機能のみが返されます。	2	タイプ 2 の特定のキーワードを含む機能のみが返されます。	3	タイプ 3 の特定のキーワードを含む機能のみが返されます。	C	現在のロケーションに定義されている機能のみが返されます (コマンド行 3 は無視されます)。	F	ローカル機能の「再帰的な」リストを呼び出します。つまり、現在のロケーション/指定したロケーションに関連するすべてのローカルコマンドが返されます。
I	内部機能番号 (IFN: Internal Function Number) も返されません。																		
T	キーワードの一部を大文字で示します (可能な省略形を示すため)。																		
S	RESULT-FIELD に返されたキーワードは列に整列されます。																		
A	内部キーワードも返されます。																		
1	タイプ 1 の特定のキーワードを含む機能のみが返されます。																		
2	タイプ 2 の特定のキーワードを含む機能のみが返されます。																		
3	タイプ 3 の特定のキーワードを含む機能のみが返されます。																		
C	現在のロケーションに定義されている機能のみが返されます (コマンド行 3 は無視されます)。																		
F	ローカル機能の「再帰的な」リストを呼び出します。つまり、現在のロケーション/指定したロケーションに関連するすべてのローカルコマンドが返されます。																		
2	<p>ローカル機能検索の開始値 (オプション)。</p> <p>キーワードは 123 の順序で指定する必要があります。</p>																		
3	<p>返されるリストに対応するロケーションを指定します。</p> <p>キーワードは 123 の順序で指定する必要があります。</p> <p>ロケーションを指定しなかった場合、コマンドプロセッサの現在のロケーションが使用されます。</p>																		
4	<p>キーワードの制限 (オプション)：</p> <p>キーワード、またはフォーマット N4 の IKN を指定した場合、このキーワードを含む機能のみが返されます。</p>																		

フィールド **RESULT-FIELD (1:n)** は、指定したリストを返します。

## IKN の HELP

指定された内部キーワード番号 (IKN) に対して、このオプションは元のキーワードを返します。

コマンド行	内容	
1	IKN で開始する必要があります。	
	オプション：	
	A	内部キーワードが示されます。
	T	キーワードの一部を大文字で示します (可能な省略形を示すため)。
2	変換する IKN をフォーマット N4 で指定します。	

フィールド **RESULT-FIELD (1)** は、キーワードを返します。

例：

入力：		出力：	
Command Line 1:	IKN	Result-Field 1:	CUSTOMER
Command Line 2:	0000002002		

## IFN の HELP

指定された内部機能番号 (IFN) に対して、このオプションは機能のキーワードを返します。

コマンド行	内容	
1	IFN で開始する必要があります。	
	オプション：	
	A	内部キーワードを含む機能は省略されません。
2	変換する IFN をフォーマット N10 で指定します。	
3	その他のオプション：	
	S	IFN に属すキーワードが <b>RESULT-FIELD (1:3)</b> に返されます。
	T	キーワードの一部を大文字で示します (可能な省略形を示すため)。
	L	IFN をロケーションとして使用した場合、IFN が返されます。
	C	IFN をコマンドとして使用した場合、IFN が返されます。

フィールド **RESULT-FIELD(1)** は、機能を返します。オプション **S** を使用した場合、機能は **RESULT-FIELD (1:3)** に返されます。

例：

入力：		出力：	
Command Line 1:	IFN	Result-Field 1:	DISPLAY INVOICE
Command Line 2:	0001048578		

## 全般的な情報の TEXT

全般的な情報については、**COMMAND-LINE (\*)**、つまりすべてのコマンド行を空白にする必要があります。**RESULT-FIELD** の最大 9 個のフィールドに次の情報が返されます。

RESULT-FIELD	内容	フォーマット
1	ユーザーテキスト用のヘッダー 1	テキスト (A40)
2	ユーザーテキスト用のヘッダー 2	テキスト (A40)
3	テキスト 「First Entry used as」	テキスト (A16)
4	テキスト 「Second Entry used as」	テキスト (A16)
5	テキスト 「Third Entry used as」	テキスト (A16)
6	エン트리 1 のキーワード数	数値 (N3)
7	エン트리 2 のキーワード数	数値 (N3)
8	エン트리 3 のキーワード数	数値 (N3)
9	カタログされた機能数	数値 (N7)

## キーワード情報の TEXT

キーワード情報については、**COMMAND-LINE (1)** には対応するキーワードが含まれている必要があります。**COMMAND-LINE (2)** にはオプションでキーワードタイプ (1、2、3、または P) を含めることができます。**COMMAND-LINE (3:6)** は空にする必要があります。

RESULT-FIELD	内容	フォーマット
1	キーワードコメントテキスト	テキスト (A40)
2	完全な長さのキーワード	テキスト (A16)
3	一意の短縮形キーワード	テキスト (A16)
4	「Keyword used as」 エントリ	テキスト (A16)
5	内部キーワード番号 (IKN)	数値 (N4)
6	キーワードの最小長	数値 (N2)
7	キーワードの最大長	数値 (N2)

RESULT-FIELD	内容	フォーマット
8	キーワードタイプ (1、2、3、1S、2S、3S、P)	テキスト (A2)

## 機能情報の TEXT

機能情報については、COMMAND-LINE (1:3) には目的のロケーションを指定するキーワードが含まれている必要があります。COMMAND-LINE (4:6) には目的の機能を指定するキーワードが含まれている必要があります。例えば、グローバルコマンド ADD USER に関する情報を返す場合、コマンド行 1、2、3、6 が空白であり、コマンド行 4 には ADD、コマンド行 5 には USER がそれぞれ含まれている必要があります。

RESULT-FIELD	内容	フォーマット
1	ランタイムアクション定義にオプション T で定義されたテキスト。	テキスト (A40)
2	指定したロケーションの内部機能番号 (IFN)。	数値 (N10)
3	指定した機能の内部機能番号 (IFN)。	数値 (N10)

## GET オプション

GET オプションは、内部コマンドプロセッサ情報や現在のコマンドプロセッサ設定をダイナミックに割り当てられたバッファ NCPWORK から読み取るために使用します。次のフィールドが使用されます。

フィールド名	内容
GETSET-FIELD-NAME (A32)	読み取る変数の名前。
GETSET-FIELD-VALUE (A32)	PROCESS COMMAND ACTION GET の実行後に指定した変数の値。

GETSET-FIELD-NAME に有効な値のリストについては、[下記](#)を参照してください。

## SET オプション

SET オプションは、バッファ NCPWORK の内部コマンドプロセッサ設定を変更するために使用します。

フィールド名	内容
GETSET-FIELD-NAME (A32)	変更する変数の名前。
GETSET-FIELD-VALUE (A32)	指定した変数に書き込まれる値。

GETSET-FIELD-NAME の設定可能値：

フィールド名	フォーマット	G/S*	内容
NAME	A8	G	現在のプロセッサの名前。
LIBRARY	A8	G	ロード元のライブラリ。
FNR	N10	G	ロード元のファイル。
DBID	N10	G	ロード元のデータベース。
TIMESTAMP	A8	G	現在のプロセッサのタイムスタンプ。
COUNTER	N10	G	アクセスカウンタ。
BUFFER-LENGTH	N10	G	NCPWORK 用に割り当てられたバイト数。
C-DELIMITER	A1	G/S	複数コマンドのデリミタ。
DATA-DELIMITER	A1	G	データの前に付けられるデリミタ。
PF-KEY	A1	G/S	PF キーをコマンドとして使用 (Y/N)。
UPPER-CASE	A1	G	大文字のキーワード (Y/N)。
UQ-KEYWORDS	A1	G	一意のキーワード (Y/N)。
IMPLICIT-KEYWORD	A1	G/S	暗黙的なキーワードエントリの識別子。
MIN-LEN	N10	G	キーワードの最小長。
MAX-LEN	N10	G	キーワードの最大長。
KEYWORD-SEQ	A8	G/S	キーワードシーケンス。
ALT-KEYWORD-SEQ	A8	G/S	代替キーワードシーケンス。
USER-SEQUENCE	A1	G	ユーザーによる KEYWORD-SEQ の上書きを許可 (Y/N)。
CURR-LOCATION	N10	G/S	現在のロケーション (IFN)。
CURR-IKN1	N10	G/S	現在のロケーションの IKN1。
CURR-IKN2	N10	G/S	現在のロケーションの IKN2。
CURR-IKN3	N10	G/S	現在のロケーションの IKN3。
CHECK-LOCATION	N10	G	最後にチェックされたロケーション (IFN)。
CHECK-IKN1	N10	G	CHECK-LOCATION の IKN1。
CHECK-IKN2	N10	G	CHECK-LOCATION の IKN2。
CHECK-IKN3	N10	G	CHECK-LOCATION の IKN3。
TOP-IKN1	N10	G	先頭キーワードの IKN1。
TOP-IKN2	N10	G	先頭キーワードの IKN2。
TOP-IKN3	N10	G	先頭キーワードの IKN3。
KEY1-TOTAL	N10	G	タイプ 1 のキーワードの数。
KEY2-TOTAL	N10	G	タイプ 2 のキーワードの数。
KEY3-TOTAL	N10	G	タイプ 3 のキーワードの数。
FUNCTIONS-TOTAL	N10	G	カタログされた機能の数。
LOCAL-GLOBAL-SEQ	A8	G/S	ローカル/グローバル機能の確認。

フィールド名	フォーマット	G/S*	内容
ERROR-HANDLER	A8	G/S	一般的なエラープログラム。
SECURITY	A1	G	Natural Security がインストール済み (Y/N)。
SEC-PREFETCH	A1	G	Natural Security データを読み取る (Y/N) または読み取り済み (D = 終了)
PREFIX1	A1	G	SYSNCP コーティリティの [Processor Header Maintenance 2] 画面のフィールド Prefix Character 1 に対応 (「キーワードエディタオプション-ヘッダー 2」セクションを参照)。
PREFIX2	A1	G	[Processor Header Maintenance 2] 画面のフィールド Prefix Character 2 に対応。
HEX1	A1	G	[Processor Header Maintenance 2] 画面のフィールド Hex. Replacement 1 に対応。
HEX2	A1	G	[Processor Header Maintenance 2] 画面のフィールド Hex. Replacement 2 に対応。
DYNAMIC	A32	G	最後のエラーメッセージのダイナミック部分 (:n:)。
LAST	-	G	データとしてスタックの先頭に置かれる最後のコマンド。
LAST-ALL	-	G	データとしてスタックの先頭に置かれる最後のコマンド群。
LAST-COM	-	G	*COM に移される最後のコマンド。
MULTI	-	G	最後の複数コマンドをデータとしてスタックの先頭に挿入。
MULTI-COM	-	G	最後の複数コマンドをシステム変数 *COM に挿入。

\*G = GET オプションで使用できます。

\*S = SET オプションで使用できます。


## USING 節

フィールドの内容 (プロセッサ名やコマンド行など) を USING 節に指定します。

USING 節では、コマンドプロセッサに送信するフィールドを指定します。


オプション	フィールド名			
	PROCESSOR-NAME	COMMAND-LINE	GETSET-FIELD-NAME	GETSET-FIELD-VALUE
CLOSE				
CHECK	必須	必須		
EXEC	必須	必須		
TEXT	必須	必須		
HELP	必須	必須		
GET	必須		必須	
SET	必須		必須	必須

## GIVING 節

 **注意:** この節はレポートモードでのみ使用できます。

GIVING 節では、各オプションの処理結果として、コマンドプロセッサによって埋められるフィールドを指定します。

オプション	フィールド名			
	NATURAL-ERROR	RETURN-CODE	RESULT-FIELD	GETSET-FIELD-VALUE
CLOSE	推奨			
CHECK	推奨	必須	必須	
EXEC	推奨	必須	必須	
TEXT	推奨	必須	必須	
HELP	推奨	必須	必須	
GET	推奨			必須
SET	推奨			

 **注意:** GIVING 節はストラクチャードモードでは利用できません。DEFINE DATA ステートメントに指定されたすべてのフィールドで構成される暗黙的な GIVING 節が存在するためです。通常、これらのフィールドはレポートモードの GIVING 節で参照されます。つまり、ストラクチャードモードでは、上記の表の「必須」でマークされたフィールドはすべて DEFINE DATA ステートメントで定義する必要があります。

## DDM : COMMAND


データ定義モジュール (DDM) COMMAND は、PROCESS COMMAND ステートメントと組み合わせて使用するために特別に作成されたものです。

```
DB:      1 File:      1 - COMMAND                      Default Sequence: ?
```

TYL	DB	NAME	F	LENG	S	D	REMARKS
1	AA	PROCESSOR-NAME	A	8	N	D DE	USING
M 1	AB	COMMAND-LINE	A	80	N	D MU/DE	USING
1	AF	GETSET-FIELD-NAME	A	32	N	D DE	USING
1	BA	NATURAL-ERROR	N	4.0	N		GIVING
1	BB	RETURN-CODE	A	4	N		GIVING
M 1	BC	RESULT-FIELD	A	80	N	MU	GIVING
1	BD	GETSET-FIELD-VALUE	A	32	N	D	USING; GIVING

\*\*\*\*\* DDM OUTPUT TERMINATED \*\*\*\*\*



 **注意:** コンパイルエラーやランタイムエラーを回避するため、使用前に DDM COMMAND という名前の DDM がタイプ C (SYSDDM メニューの DDM Type フィールド) でカタログされていることを確認してください。DDM を再カタログする場合、SYSDDM ユーティリティの DBID/FNR 指定は無視されます。

DDM COMMAND のフィールドは次のとおりです。

DDM フィールド	説明
PROCESSOR-NAME	PROCESS COMMAND ステートメントが発行されるコマンドプロセッサの名前です。指定するコマンドプロセッサはカタログする必要があります。
COMMAND-LINE	コマンドプロセッサで処理されるコマンド行 (オプション CHECK、EXEC)、またはプログラムに返されるユーザーテキストやヘルプテキストに対するキーワード/コマンド (オプション TEXT、HELP) です。このフィールドは複数行指定できます。
GETSET-FIELD-NAME	GET および SET オプションとともに使用し、読み取り (GET) または書き込み (SET) が行う定数や変数の名前を指定します。
RETURN-CODE	このフィールドには、ランタイムアクション定義内で指定されたオプション EXEC や CHECK による処理のリターンコードが入ります (Natural SYSNCP ユーティリティを参照)。
NATURAL-ERROR	すべてのオプションと組み合わせて使用します。DEFINE DATA でこのフィールドを使用した場合、コマンドプロセッサに対する Natural エラーコードが返されます。フィールドが存在しない場合、エラーの発生時に Natural ランタイムエラー処理が開始されます。
RESULT-FIELD	このフィールドには、ランタイムアクション定義内で指定された各種オプションの使用結果情報が入ります (Natural SYSNCP ユーティリティのランタイムアクションを参照)。このフィールドは複数行指定できます。
GETSET-FIELD-VALUE	GET および SET オプションとともに使用し、GETSET-FIELD-NAME フィールドで指定された定数や変数の値が入ります。

## 例

### ■ 例 1 - PROCESS COMMAND ACTION CLOSE

- 例 - PROCESS COMMAND ACTION EXEC2

### 例 1 - PROCESS COMMAND ACTION CLOSE

```

/* EXAM-CLS - Example for PROCESS COMMAND ACTION CLOSE (Structured Mode)
/*****
DEFINE DATA LOCAL
  01 COMMAND VIEW OF COMMAND
END-DEFINE
/*
PROCESS COMMAND ACTION CLOSE
/*
DEFINE WINDOW CLS
INPUT WINDOW = 'CLS'
  'NCPWORK has just been released.'
/*
END

```

### 例 - PROCESS COMMAND ACTION EXEC2

```

/* EXAM-EXS - Example for PROCESS COMMAND ACTION EXEC (Structured Mode)
/*****
DEFINE DATA LOCAL
  01 COMMAND VIEW OF COMMAND
    02 PROCESSOR-NAME
    02 COMMAND-LINE (1)
    02 NATURAL-ERROR
    02 RETURN-CODE
    02 RESULT-FIELD (1)
  01 MSG (A65) INIT <'Please enter a command.'>
END-DEFINE
/*
REPEAT
  INPUT (AD=MIT' ' IP=OFF) WITH TEXT MSG
    'Example for PROCESS COMMAND ACTION EXEC (Structured Mode)' (I)
  / 'Command ==>' COMMAND-LINE (1) (AL=64)
  /*****
  PROCESS COMMAND ACTION EXEC
  USING
    PROCESSOR-NAME = 'DEMO'
    COMMAND-LINE (1) = COMMAND-LINE (1)
  /*****
  COMPRESS 'NATURAL-ERROR =' NATURAL-ERROR TO MSG
END-REPEAT
END

```



**注意:** 他のプログラム例については、SYSNCP ライブラリを参照してください。これらのプログラムはすべて EXAM で始まります。

# 98

## PROCESS PAGE

---

■ 機能 .....	658
■ 構文 1 - PROCESS PAGE .....	658
■ 構文 2 - PROCESS PAGE USING .....	660
■ 構文 3 - PROCESS PAGE UPDATE .....	662
■ 構文 4 - PROCESS PAGE MODAL .....	665

## 機能

PROCESS PAGE ステートメントは、外部レンダリングエンジン（Application Designer など）に一般的なインターフェイスの説明を提供し、Natural 内部データ表現と外部データ表現をリンクします。このリンクを介して、データおよびイベントがブラウザベースの外部アプリケーションとやりとりされます。ただし、レンダリング情報はやりとりされません。

詳細については、『Natural for Ajax』ドキュメントを参照してください。

## 構文 1 - PROCESS PAGE


```
PROCESS PAGE operand1
[WITH PARAMETERS
  {[NAME] operand2 [VALUE] operand3 [ (EM=em value) ]}...
END-PARAMETERS]
[GIVING operand9]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[対話型処理用の画面生成](#)」

### 構文説明 - 構文 1

PROCESS PAGE ステートメントの構文 1 は、通常 Natural アダプタ内でのみ使用されます。アダプタとは、Natural アプリケーションと Web ページ間のインターフェイスを形成する Natural オブジェクトです。アダプタは、レイアウトの保存時に Application Designer によって自動的に作成／更新されます。

 **注意:**

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
operand1	C S	A U	可	不可
operand2	C S	A U	可	不可
operand3	C S A	A U N P I F B D T L	可	可
operand9	S	I4	可	可

## 構文要素の説明：

<i>operand1</i>	外部ページレイアウトの名前です。
<i>operand2</i>	<i>operand3</i> の相互転送に使用される外部データフィールドの名前が入ります。
<i>operand3</i>	転送される Natural データフィールドの名前が入ります。
<b>GIVING</b> <i>operand9</i>	<b>GIVING</b> 節： 要求を実行できなかった場合は、 <i>operand9</i> に Natural エラーが含まれます。
<b>EM=</b>	データ転送時に使用される編集マスク。  詳細については、『パラメータリファレンス』のセッションパラメータ EM を参照してください。

## Application Designer によって作成されたアダプタの例：

```

* PAGE1: PROTOTYPE      --- CREATED BY Application Designer ---
* PROCESS PAGE USING 'XXXXXXX' WITH
* INFOPAGENAME RESULT YOURNAME
DEFINE DATA PARAMETER
1 INFOPAGENAME (U) DYNAMIC
1 RESULT (U) DYNAMIC
1 YOURNAME (U) DYNAMIC
END-DEFINE
*
PROCESS PAGE U'/njxdemos/helloworld' WITH
PARAMETERS
  NAME U'infopagename'
  VALUE INFOPAGENAME
  NAME U'result'
  VALUE RESULT
  NAME U'yourname'
  VALUE YOURNAME
END-PARAMETERS
*
* TODO: Copy to your calling program and implement.
/*/*( DEFINE EVENT HANDLER
* DECIDE ON FIRST *PAGE-EVENT
* VALUE U'nat:page.end'
* /* Page closed.
* IGNORE
* VALUE U'onHelloWorld'
* /* TODO: Implement event code.
* PROCESS PAGE UPDATE FULL
* NONE VALUE
* /* Unhandled events.
* PROCESS PAGE UPDATE
* END-DECIDE

```

```

/*/*) END-HANDLER
*
END
    
```

## 構文 2 - PROCESS PAGE USING

```

PROCESS PAGE USING operand4
    [ { WITH {operand5} ... } ]
    [GIVINGoperand9]
    
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[対話型処理用の画面生成](#)」

### 構文説明 - 構文 2

この構文は、ページレイアウト（Application Designer または類似のツールで作成）から生成されたアダプタタイプのオブジェクトを使用してリッチ GUI 入力／出力処理を実行するために使用します。

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
operand4	C S	A	可	不可
operand5	S A G	A U N P I F B D T L	可	可
operand9	S	I4	可	可

構文要素の説明：

<b>USING</b>	アダプタ名：
<i>operand4</i>	<p>Natural システムファイルに事前に保存されているアダプタ定義を呼び出します。『プログラミングガイド』の「<a href="#">リッチ GUI ページの処理 - アダプタ</a>」も参照してください。</p> <p>アダプタ名 (<i>operand4</i>) には、1～8 文字の英数字定数またはユーザー定義変数を使用できます。変数を使用する場合、事前の定義が必要です。</p> <p>アダプタ名にはアンパサンド (&amp;) を含めることができます。この文字は実行時に Natural システム変数 *LANGUAGE の現在の値で置き換えられます。この機能は履歴上の理由で用意されています。複数言語のアダプタが必要な場合、外部レンダリングシステム（例えば、Application Designer）の機能を使用してください。</p>

	<p><b>注意:</b> 新しいアプリケーションでは、複数言語にするためのアンパサンド機能は必要ありません。Application Designer などを使用して設計されたページでは、レイアウト設計の一部として複数言語の情報を保持できます。『Natural for Ajax』ドキュメントの「Multi Language Management」を参照してください。</p>
<i>operand5</i>	<p>フィールド指定：</p> <p>データベースフィールドまたはユーザー定義変数（すべて事前の定義が必要）のリストを指定します。フィールドは、番号、シーケンス、フォーマット、長さ、オカレンス数（配列の場合）について、参照されているアダプタに準拠している必要があります。そうでない場合、エラーが発生します。</p> <p>データベースフィールドの内容が PROCESS PAGE 処理の結果として変更される場合は、データエリアにある値のみが変更されます。データベースの内容を変更するためには、適切なデータベース UPDATE/STORE ステートメントを使用する必要があります。</p> <p>「プログラムで定義されたフィールドに基づく PROCESS PAGE USING」を参照してください。</p>
NO PARAMETER	「パラメータリストなしの PROCESS PAGE USING」を参照してください。
GIVING <i>operand9</i>	<p><b>GIVING 節：</b></p> <p>要求を実行できなかった場合は、<i>operand9</i> に Natural エラーが含まれます。</p> <p><b>注意:</b> GIVING 節では、アダプタオブジェクトの起動中または実行中にエラーが発生した場合に共通 Natural エラー処理を中断します。Natural モジュールを後方追跡して ON ERROR 節を見つける代わりに、Natural エラーコードが変数 (<i>operand9</i>) に渡され、次のステートメントの実行が続行されます。</p>

### パラメータリストなしの PROCESS PAGE USING

PROCESS PAGE USING をパラメータリストなしで使用する場合、次の要件を満たす必要があります。

- アダプタ名 (*operand5*) は、英数字定数（最大 8 文字）として指定する必要があります。
- この方法で使用されるアダプタは、そのアダプタを参照するプログラムのコンパイル前に事前に作成されている必要があります。
- 処理対象のフィールドの名前は、コンパイル時にアダプタソース定義から動的に取得される必要があります。プログラムとアダプタの両方で使用されるフィールド名が同一である必要があります。
- PROCESS PAGE ステートメントで参照されるすべてのフィールドにその時点でアクセスできる必要があります。
- ストラクチャードモードでは、フィールドが事前に定義されている必要があります（処理ループまたはビューでデータベースフィールドが適切に参照されている必要があります）。
- ページレイアウトが変更されたとき、アダプタを使用しているプログラムの再カタログは不要です。ただし、配列構造や名前、フィールドのフォーマット/長さが変更されたとき、また

はアダプタでフィールドが追加／削除されたときは、アダプタを使用しているプログラムの再カタログが必要になります。

- プログラムのコンパイル時にアダプタソースを利用できる必要があります。利用できない場合、PROCESS PAGE USING ステートメントはコンパイルできません。



**注意:** アダプタを利用できない場合でもプログラムをコンパイルする必要がある場合は、NO PARAMETER を指定してください。これにより、アダプタを利用できない場合でもPROCESS PAGE USING ステートメントのコンパイルが可能になります。

### プログラムで定義されたフィールドに基づく PROCESS PAGE USING

プログラム (*operand5*) 内で処理するフィールドの名前を指定することで、プログラム内のフィールドとアダプタ内のフィールドに異なる名前を使用できます。

プログラム内のフィールドのシーケンスは、アダプタ内のフィールドのシーケンスと一致している必要があります。Natural マップをアダプタオブジェクトとして使用する場合は、マップエディタによって、マップで指定されたフィールドがフィールド名でアルファベット順にソートされることに留意してください。詳細については、『エディタ』ドキュメントのマップエディタの説明を参照してください。

プログラムエディタの行コマンド `.I(adaptername)` を使用して、指定したアダプタで定義されたフィールドに基づくパラメータリスト付きの完全な PROCESS PAGE USING ステートメントを取得できます。

アダプタのレイアウトが変更されたとき、アダプタを使用しているプログラムの再カタログは不要です。ただし、アダプタ内のフィールド名、フィールドのフォーマット／長さ、または配列構造が変更されたとき、またはアダプタでフィールドが追加／削除されたときは、プログラムの再カタログが必要になります。

実行時に、プログラムで指定されたフィールドのフォーマットと長さがアダプタで指定されたフィールドと一致しているかどうかチェックされます。両方のレイアウトが一致しない場合は、エラーメッセージが生成されます。

## 構文 3 - PROCESS PAGE UPDATE

```
PROCESS PAGE UPDATE [FULL] [event-option]  
[GIVINGoperand9]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[対話型処理用の画面生成](#)」



## 構文説明 - 構文 3

PROCESS PAGE UPDATE ステートメントは、PROCESS PAGE ステートメントに戻って再実行するために使用します。通常、前の PROCESS PAGE ステートメントのデータ入力処理が不完全だったイベント処理から戻るために使用します。



**注意:** PROCESS PAGE ステートメントとそれに対応する PROCESS PAGE UPDATE ステートメントの間に、INPUT、WRITE、PRINT、または DISPLAY ステートメントを実行することはできません。

PROCESS PAGE UPDATE ステートメントが実行されると、PROCESS PAGE ステートメントが実行されたときに存在していたサブルーチン、特殊条件、およびループ処理を考慮してプログラムステータスが再位置づけされます（PROCESS PAGE ステートメントのステータスがまだ有効な場合）。PROCESS PAGE ステートメントの実行後にループが開始され、PROCESS PAGE UPDATE ステートメントがそのループ内にある場合、ループは中断され、PROCESS PAGE UPDATE の結果として再処理された PROCESS PAGE ステートメントの後で再開されます。

PROCESS PAGE ステートメントの実行後にサブルーチン階層が呼び出され、サブルーチン内で PROCESS PAGE UPDATE が実行された場合、Natural は自動的に全サブルーチンをさかのぼり、PROCESS PAGE ステートメントの位置にプログラムステータスを再位置づけします。

PROCESS PAGE ステートメントが実行されたステータスがすでに終了してしまった時点では、PROCESS PAGE ステートメントをループ、サブルーチン、または特殊条件ブロック内に位置づけ、PROCESS PAGE UPDATE ステートメントを実行することはできません。このエラー条件が検出されると、エラーメッセージが生成され、プログラムの実行が終了します。

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand9</i>	S	I4	可	可

構文要素の説明：

<b>FULL</b>	<p>PROCESS PAGE UPDATE ステートメントに FULL オプションを指定すると、対応する PROCESS PAGE ステートメントが完全に再実行されます。</p> <ul style="list-style-type: none"> <li>■ 通常の PROCESS PAGE UPDATE ステートメント（FULL オプションなし）では、PROCESS PAGE ステートメントと PROCESS PAGE UPDATE ステートメントの間で変更された変数の内容は表示されません。つまり、画面上の変数の内容はすべて PROCESS PAGE ステートメントが実行された時点のものになります。</li> <li>■ PROCESS PAGE UPDATE FULL ステートメントでは、最初の PROCESS PAGE ステートメントの実行後に行われたすべての変更は、PROCESS PAGE ステートメントが再実行されたときに適用されます。つまり、画面上の変数の値はすべて PROCESS PAGE UPDATE ステートメントが実行された時点のものになります。</li> </ul>
-------------	---

<i>event-option</i>	<p><b>EVENT</b> オプション： 下記の「<b>EVENT</b> オプション」を参照してください。</p>
<b>GIVING</b> ( <i>operand9</i> )	<p><b>GIVING</b> 節： 要求を実行できなかった場合は、<i>operand9</i>に <b>Natural</b> エラーが含まれます。</p>

ユーザープログラムフラグメントの例：

```
PROCESS PAGE USING "HELLOW-A"
*
/*( DEFINE EVENT HANDLER
DECIDE ON FIRST *PAGE-EVENT
VALUE U'nat:page.end'
/* Page closed.
IGNORE
VALUE U'onHelloWorld'
COMPRESS "HELLO WORLD" YOURNAME INTO RESULT
PROCESS PAGE UPDATE FULL
NONE VALUE
/* Unhandled events.
PROCESS PAGE UPDATE
END-DECIDE
/*) END-HANDLER
```

### EVENT オプション

```
AND SEND EVENT operand6
[WITH PARAMETERS
{[NAME] operand7 [VALUE] operand8 [ { (EM=value) } ]...
END-PARAMETERS]
```

このオプションでは、外部 I/O システムに特定の機能を実行するように指示できます。これらの機能は、外部 I/O システムの一部であるか、またはフォーカスの設定、メッセージボックスの表示などの出力処理に関する特別な機能を実装します。

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand6</i>	C S	A U	可	不可
<i>operand7</i>	C S	A U	可	不可
<i>operand8</i>	C S A	A U N P I F B D T L	可	可

構文要素の説明：

<i>operand6</i>	外部 I/O システムから要求されるイベント：  外部 I/O システムの実装に応じて、イベントを使用できます。『 <i>Natural for Ajax</i> 』ドキュメントの「 <i>Sending Events to the User Interface</i> 」を参照してください。
<i>operand7</i>	外部データフィールド名：  <i>operand7</i> には、 <i>operand8</i> の相互転送に使用される外部データフィールドの名前が入りません。
<i>operand8</i>	<b>Natural</b> データフィールド：  <i>operand8</i> には、転送される Natural データフィールドが入ります。
<b>EM=</b>	編集マスク：  データ転送時に使用される編集マスク。  編集マスクの詳細については、『 <i>パラメータリファレンス</i> 』でセッションパラメータ EM を参照してください。

## 構文 4 - PROCESS PAGE MODAL

```
PROCESS PAGE MODAL
  statement ...
END-PROCESS
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ESCAPE](#) | [PROCESS PAGE](#)

関連機能グループ：

- ループ実行
- 対話型処理用の画面生成

## 構文説明 - 構文 4

PROCESS PAGE MODAL ステートメントは、処理ブロックを開始し、モーダルリッチ GUI ウィンドウの存続期間を制御するために使用します。



**注意:** PROCESS PAGE MODAL ステートメントはバッチモードでは使用できません。

PROCESS PAGE MODAL ステートメントブロックに入ると、まだ表示されていないレポート 0 のデータが最初に表示されます。

システム変数 \*PAGE-LEVEL が増加し、モーダルページのオープン準備が行われます。モーダルページの実際のオープンは次の PROCESS PAGE USING 'adapter' WITH ステートメントで実行されます。



**注意:** PROCESS PAGE MODAL ステートメントとそれに対応する END-PROCESS ステートメントの間に、レポート 0 を参照する PRINT、WRITE、INPUT、または DISPLAY ステートメントを実行することはできません。

PROCESS PAGE MODAL ステートメントブロックを抜けると、次のアクションが実行されます。

- モーダルページがこのレベルでオープンされている場合、モーダルページがクローズされます。
- システム変数 \*PAGE-LEVEL が減少し、システム変数 \*PAGE-EVENT が、ステートメントブロックに入る前の値に戻されます。

構文要素の説明：

<i>statement</i>	状況に応じて、 <i>statement</i> の代わりに、1つ以上の適切なステートメントを指定する必要があります。不要なステートメントがある場合は、IGNORE ステートメントを挿入します。
<b>END-PROCESS</b>	PROCESS PAGE MODAL ステートメントを終了するには、Natural 予約語 END-PROCESS を使用する必要があります。

例：

```
* Name: First Demo/Open modal!
*
PROCESS PAGE USING "EMPTY-A"
*
/*( DEFINE EVENT HANDLER
DECIDE ON FIRST *PAGE-EVENT
  VALUE U'nat:page.end', U'onClose'
  /* Page closed.
  IGNORE
  VALUE U'onNextLevel'
  PROCESS PAGE MODAL
  FETCH RETURN "EMPTY-P"
```

```
END-PROCESS
  PROCESS PAGE UPDATE
  NONE VALUE
  PROCESS PAGE UPDATE
END-DECIDE
/*) END-HANDLER
END
```

---

# 99 PROPERTY

---

▪ 機能 .....	670
▪ 構文説明 .....	670
▪ 例 .....	671

```
PROPERTY property-name
  OF [INTERFACE] interface-name
  IS operand
END-PROPERTY
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CREATE OBJECT](#) | [DEFINE CLASS](#) | [INTERFACE](#) | [METHOD](#) | [SEND METHOD](#)

関連機能グループ：「[コンポーネントベースプログラミング](#)」

## 機能

PROPERTY ステートメントでは、インターフェイス定義外で、実装としてオブジェクトデータ変数オペランドをプロパティに割り当てます。

問題のインターフェイス定義がコピーコードから取得され、クラス固有の方法で実装される場合は、これが使用されます。

[DEFINE CLASS](#) ステートメント内で、インターフェイス定義の後にのみ使用できます。

指定するインターフェイス名およびプロパティ名は、[DEFINE CLASS](#) ステートメントの [INTERFACE](#) 節で定義されている必要があります。

## 構文説明

<i>property-name</i>	これは、 <a href="#">プロパティ</a> に割り当てられた名前です。
OF <i>interface-name</i>	これは、 <a href="#">インターフェイス</a> に割り当てられた名前です。
IS <i>operand</i>	IS 節の <i>operand</i> により、オブジェクトデータ変数をプロパティ値の保存場所として割り当てます。
END-PROPERTY	PROPERTY ステートメントを終了するには、Natural 予約語 END-PROPERTY を使用する必要があります。



## 例

---

**METHOD** ステートメントのドキュメントに記載の例は、同じインターフェイスを2つのクラスに別々に実装する方法と、これを実現するために **PROPERTY** ステートメントと **METHOD** ステートメントを使用する方法を示しています。



# 100 READ

---

■ 機能 .....	674
■ 構文説明 .....	675
■ READ で使用可能なシステム変数 .....	684
■ 例 .....	685

{	READ	}	[[	ALL	]]	[ <i>MULTI-FETCH-clause</i> ]
	BROWSE			( <i>operand1</i> )		[RECORDS][IN][FILE]
						<i>view-name</i>
						[PASSWORD= <i>operand2</i> ]
						[CIPHER= <i>operand3</i> ]
						[WITH REPOSITION]
						[ <i>sequence/range-specification</i> ]
						[STARTING WITH ISN= <i>operand4</i> ]
						[WHERE <i>logical-condition</i> ]
						<i>statement ...</i>
END-READ						(ストラクチャードモードのみ)
[LOOP]						(レポートモードのみ)

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [GET TRANSACTION DATA](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [HISTOGRAM](#) | [GET](#) | [GET SAME](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [RETRY](#) | [STORE](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

READ ステートメントは、データベースからレコードを読み取るために使用します。レコードは物理順、Adabas ISN 順、またはディスクリプタ（キー）フィールドの値順で取得できます。

このステートメントでは、処理ループを開始します。

「[READ ステートメント](#)」（『[プログラミングガイド](#)』）も参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	N P I B*	可	不可
<i>operand2</i>	C S	A	可	不可
<i>operand3</i>	C S	N	可	不可
<i>operand4</i>	C S	N P I B*	可	不可

\* *operand1* および *operand4* のフォーマット B は、4 以下の長さでのみ使用できます。

構文要素の説明：

<i>operand1</i>	<p>読み取るレコード数：</p> <p>読み取るレコード数は、<i>operand1</i> を（キーワード READ の直後にカッコで囲んで）指定することで制限できます。この際、キーワード READ の直後にカッコで囲んで数値定数（0～4294967295）または変数として指定します。次に例を示します。</p> <pre>READ (5) IN EMPLOYEES ...  MOVE 10 TO CNT(N2) READ (CNT) EMPLOYEES ...</pre> <p>このステートメントでは、指定した制限が、LIMIT ステートメントで設定した制限より優先されます。</p> <p>プロファイルまたはセッションパラメータ LT で設定された制限のほうが小さい場合、LT 制限が適用されます。</p> <p><b>注意:</b></p> <ol style="list-style-type: none"> <li>読み取るレコード数が4桁の場合、先頭にゼロを付けて指定します（0nnnn）。カッコで囲まれた4桁の数はすべてステートメントの行番号参照として Natural で解釈されるためです。</li> <li><i>operand1</i> は、READ ループに入るときに評価されます。<i>operand1</i> の値が READ ループ内で変更されても、読み取られるレコード数には影響しません。</li> </ol>
ALL	<p>全レコードを読み取ることを強調するために、オプションでキーワード ALL を指定できます。</p>

<i>MULTI-FETCH-clause</i>	以下の「 <a href="#">MULTI-FETCH 節</a> 」を参照してください。
<i>view-name</i>	<p>ビュー名：</p> <p><i>view-name</i> として、ビューの名前を指定します。この名前は、<code>DEFINE DATA</code> ステートメント内、あるいはプログラム外のグローバルデータエリアまたはローカルデータエリアに事前に定義しておく必要があります。</p> <p>レポーティングモードでは、<i>view-name</i> を DDM の名前にすることもできます。</p>
<b>PASSWORD</b> <b>CIPHER</b>	<p><b>PASSWORD</b> および <b>CYPHER</b> 節：</p> <p>これらの節は Adabas または VSAM データベースにのみ適用可能です。Entire System Server では使用できません。</p> <p><b>PASSWORD</b> 節は、パスワード保護されているファイルからデータを取得するときのパスワードの指定に使用します。</p> <p><b>CIPHER</b> 節は、暗号化されているファイルからデータを取得するときのサイファキーの指定に使用します。</p> <p>詳細については、ステートメント <b>FIND</b> および <b>PASSW</b> を参照してください。</p>
<b>WITH REPOSITION</b>	このオプションは、 <code>READ</code> ステートメントを再位置決めイベントに反応するように設定するために使用します。「 <a href="#">WITH REPOSITION オプション</a> 」を参照してください。
<i>sequence/range-specification</i>	このオプションでは、取得順序／範囲を指定します。「 <a href="#">順序／範囲の指定</a> 」を参照してください。
<b>STARTING WITH ISN=operand4</b>	<p>この節は Adabas および VSAM データベースにのみ適用されます。</p> <p><b>Adabas</b> へのアクセス</p> <p>この節は、物理順または論理順（昇順／降順）の <code>READ</code> ステートメントと組み合わせて使用します。指定された値 (<i>operand4</i>) は Adabas ISN（内部シーケンス番号）を表し、<code>READ</code> ループを開始するレコードを明確に指定するために使用します。</p> <p>■ 論理順</p> <p>等号 (=) で記述されても、<code>READ</code> ステートメントは対応するディスクリプタフィールドに同じ開始値を持つレコードのみを返すわけではなく、指定された開始値で始めて昇順または降順に論理的な参照を開始します。いくつかのレコードがディスクリプタフィールドに同じ内容を持っている場合、それらは ISN ソート順に戻されます。</p> <p><code>STARTING WITH ISN</code> 節は、開始値が最初のレコードのディスクリプタ値と一致している場合にのみ適用される一種の「第2レベル」の選択条件になります。開始値と同じディスクリプタ値の全レコードおよび開始 ISN 「以下」（降順 <code>READ</code> については「以</p>

	<p>上」) の ISN は Adabas で無視されます。READ ループで返される最初のレコードは次のいずれかです。</p> <ul style="list-style-type: none"> <li>■ ディスクリプタ = 開始値の最初のレコードおよび開始 ISN より「高い」(降順 READ については「低い」) ISN。</li> <li>■ このようなレコードが存在しない場合は、開始値より「大きい」(降順 READ については「小さい」) ディスクリプタを持つ最初のレコード。</li> </ul> <p>■ 物理順 レコードは、それらが物理的に保存されている順序で返されます。STARTING WITH ISN 節を指定した場合、開始 ISN と同じ ISN のレコードに達するまで、Adabas では全レコードが無視されます。返される最初のレコードは、ISN = 開始 ISN のレコードに続く次のレコードです。</p> <p><b>VSAM へのアクセス</b></p> <p>この節は物理順でのみ使用できます。指定される値 (<i>operand4</i>) は、VSAMRBA (ESDSの相対バイトアドレス) または RRN (RRDSの相対レコード番号) を表し、読み取り操作の開始値として使用されます。</p> <p>例</p> <p>この節は、処理を継続する次のレコードを簡単に指定するために、処理が中断された READ ループ内での再位置決めで使用できます。これは特に、次のレコードをディスクリプタ値で一意に識別できない場合に役立ちます。また、分散クライアント/サーバーアプリケーションで、レコードをサーバープログラムで読み取ってクライアントプログラムで処理し、レコードを順次処理ではなくバッチ処理する場合にも役立ちます。</p> <p>例については、下記のプログラム <b>REASISND</b> を参照してください。</p>
<b>WHERE</b> <i>logical-condition</i>	「 <b>WHERE 節</b> 」を参照してください。
<b>END-READ</b>	READ ステートメントを終了するには、Natural 予約キーワード <b>END-READ</b> を使用する必要があります。

## MULTI-FETCH 節



**注意:** この節は、Adabas または DB2 データベースでのみ使用できます。

MULTI-FETCH { <ul style="list-style-type: none"> <li>ON</li> <li>OFF</li> <li>OF <i>multi-fetch-factor</i></li> </ul>
---

詳細については、『プログラミングガイド』の「MULTI-FETCH節」(Adabas)、または『データベース管理システムインターフェイス』ドキュメントの「Natural for DB2」で「複数行の処理」(SQL)を参照してください。

## WITH REPOSITION オプション



**注意:** このオプションは、基準データベースが Adabas、VSAM、または DL/I の場合にのみ適用されます。

WITH REPOSITION オプションでは、READ ステートメントを再位置決めイベントに反応するように設定できます。これにより、アクティブな READ ループ内で別の開始値に再位置決めを行うことができます。READ ステートメントの処理は新しい開始値で続きます。

READ ステートメントを WITH REPOSITION オプション付きで使用すると、再位置決めイベントが 2 つの方法のいずれかで起動されます。

1. **ESCAPE TOP REPOSITION** ステートメントが実行される時。ESCAPE TOP REPOSITION ステートメントの実行時、Natural はループ開始への即時分岐を行い、再スタートを実行します。つまり、データベースは、検索値変数の現在の内容に従って、ファイルの新しいレコードに再位置決めを行います。同時に、ループカウンタ \*COUNTER をゼロにリセットします。
2. READ ループがデータベースから次のレコードを取り出そうとして、システム変数 \*COUNTER の値が 0 であるとき。



**注意:** アクティブな READ ループ内で \*COUNTER が 0 に設定される場合、現在のレコードの処理は続きます。ループ開始への即時分岐は行われません。

## 機能上の考慮事項

- READ ステートメントにループ制限 (例: READ (10) EMPLOYEES WITH REPOSITION ...) があり、再スタートイベントが起動された場合、再位置決めが発生するまでにすでに処理されたレコード件数にかかわらず、ループは別の新しい 10 レコードを取得します。
- **ESCAPE TOP REPOSITION** ステートメントが実行される場合、ただし、(WITH REPOSITION キーワードが READ ステートメントに設定されていないか、事後ループステートメントが READ 以外であるために) 最も内側のループが再位置決めできないときは、対応するランタイムエラーが発行されます。
- ESCAPE TOP ステートメントは参照を許可しないので、最も内側の処理ループが READ ..WITH REPOSITION ステートメントであれば、再位置決めイベントの開始だけができます。
- 再位置決めイベントは **AT START OF DATA** セクションの実行を起動せず、ループ制限オペランド (変数の場合) の再評価も起動しません。



- 検索値が変更されなかった場合、ループは最初のループ開始のように同じレコードに再位置決めします。

## 順序／範囲の指定

取得順序／範囲の指定には、3つの構文オプションを使用できます。

構文オプション1：


[IN] [PHYSICAL]	ASCENDING DESCENDING VARIABLE <i>operand5</i> DYNAMIC <i>operand5</i>	[SEQUENCE]
-----------------	--	------------

構文オプション2：

{ BY WITH }	ISN	= EQ EQUAL TO [STARTING] FROM	<i>operand6</i>	{ THRU ENDING AT }	<i>operand7</i>
----------------	-----	--	-----------------	-----------------------	-----------------

構文オプション3：

[IN] [LOGICAL]	ASCENDING DESCENDING VARIABLE <i>operand5</i> DYNAMIC <i>operand5</i>	[SEQUENCE]	{ BY WITH }	<i>descriptor</i>
-------------------	--	------------	----------------	-------------------



**注意:**

1. Entire System Server で、構文オプション [2] と [3] は使用できません。
2. 図3の比較演算子を使用した場合は、オプション ENDING AT、THRU、および TO は使用できません。これらの比較演算子は HISTOGRAM ステートメントにも有効です。

オペランド定義テーブル：

オペランド	構文要素	フォーマット										ステートメント参照	ダイナミック定義	
operand5	S	A											可	不可
operand6	C S	A	N	P	I	F	B*	D	T	L			可	不可
operand7	C S	A	N	P	I	F	B*	D	T	L			可	不可

\* operand6 および operand7 のフォーマット B は、4 以下の長さでのみ使用できます。

## 構文要素の説明：

<b>READ IN PHYSICAL SEQUENCE</b>	<p>PHYSICAL SEQUENCE は、データベースに物理的に保存されている順にレコードを読み取る場合に使用します。</p> <p><b>注意:</b> VSAM データベースの場合：READ PHYSICAL は ESDS と RRDS にのみ適用されません。</p> <p>PHYSICAL はデフォルトの順序です。</p>
<b>READ BY ISN</b>	<p>READ BY ISN は、Adabas および VSAM データベースに対してのみ使用できます。VSAM データベースに対しては ESDS および RRDS にのみ有効です。</p> <p>READ BY ISN は、レコードを Adabas ISN（内部シーケンス番号）順に読み取る際に使用します。または、VSAMRBA（ESDSの相対バイトアドレス）やRRN（RRDSの相対レコード番号）の順に読み取ります。</p> <p><b>注意:</b> XML データベースの場合：READ BY ISN は Tamino オブジェクト ID の順に従って XML オブジェクトを読み取るために使用します。</p>
<b>READ IN LOGICAL SEQUENCE</b>	<p>LOGICAL SEQUENCE は、レコードをディスクリプタ（キー）フィールドの値の順に読み取るために使用します。</p> <p>ディスクリプタが指定されると、レコードはそのディスクリプタの値の順序で読み取られます。順序制御には、ディスクリプタ、サブディスクリプタ、スーパーディスクリプタ、またはハイパーディスクリプタを使用できます。フォネティックディスクリプタ、ピリオディックグループ内のディスクリプタ、またはピリオディックグループフィールドを含むスーパーディスクリプタは使用できません。</p> <p>ディスクリプタを指定しなかった場合、DDM に指定されたデフォルトのディスクリプタ（Default Sequence フィールド）が使用されます。</p> <p><b>注意:</b></p> <ol style="list-style-type: none"> <li>DL/I データベースの場合：ディスクリプタはルートセグメントのシーケンスフィールドまたは 2 次インデックスフィールドにする必要があります。2 次インデックスフィールドを指定する場合、PCB の PROCSEQ パラメータにも指定する必要があります。Natural は、この PCB とそれに対応する階層構造によってデータベースを処理します。HDAM データベースはルートセグメントを検索するためにランダムルーチンを使用するため、HDAM データベースに対して READ LOGICAL を使用しても有効な結果は返されません。したがって、HDAM データベースには READ PHYSICAL を使用してください。</li> <li>VSAM データベースの場合：LOGICAL はプライマリおよびオルタネートキー定義のある KSDS、またはオルタネートキー定義のある ESDS にのみ有効です。</li> </ol> <p>順序制御に使用するディスクリプタが空値省略で定義されている場合（Adabas のみ）、ディスクリプタ値が空値であるレコードは読み取られません。</p> <p>順序制御に使用するディスクリプタがマルチプルバリューフィールドの場合（Adabas のみ）、存在する値の個数に応じて同じレコードが何回も読み取られます。</p>

	<p><b>注意:</b> READ IN LOGICAL SEQUENCE については、『プログラミングガイド』でも説明されています。「データベースアクセスのためのステートメント」の「READ ステートメント」を参照してください。</p>
<p><b>ASCENDING   DESCENDING   VARIABLE   DYNAMIC SEQUENCE</b></p>	<p>この節は、Adabas、XML データベース、VSAM、および SQL データベースにのみ適用されます。READ PHYSICAL ステートメントでは、VSAM および DB2 データベースにのみ適用できます。</p> <p>この節を使用して、レコードを昇順または降順のどちらで読み取るかを指定できます。</p> <ul style="list-style-type: none"> <li>■ デフォルトは昇順です。昇順はキーワード ASCENDING を使用して明示的に指定することもできます。</li> <li>■ レコードを降順で読み取る場合は、キーワード DESCENDING を指定します。</li> <li>■ レコードを昇順または降順のどちらで読み取るかを事前に決めずに実行時に決めるには、変数 (<i>operand5</i>) の前に VARIABLE キーワードまたは DYNAMIC キーワードを指定します。<i>Operand5</i> は、フォーマット/長さ A1 で、値 A (「昇順」) または D (「降順」) が含まれている必要があります。             <ul style="list-style-type: none"> <li>■ キーワード VARIABLE を使用した場合、読み取り順 (<i>operand5</i> の値) は、READ 処理ループの開始時に評価され、READ ループ内で <i>operand5</i> フィールドが変更されるかどうかに関係なく、ループが終了するまで変わりません。</li> <li>■ キーワード DYNAMIC を使用した場合、読み取り順 (<i>operand5</i> の値) は、READ 処理ループでの各レコード取り出し前に評価され、レコードからレコードへ変更できます。これは、READ ループの任意の場所で昇順から降順 (または、逆) へのスクロール順の変更を可能にします。</li> </ul> </li> </ul> <p><b>注意:</b></p> <ol style="list-style-type: none"> <li>1. Adabas データベースの場合：DYNAMIC 順を使用するには、システムで Adabas V7 (またはそれ以降) が必要です。</li> </ol>
<p><b>STARTING FROM ... ENDING AT/TO</b></p>	<p>STARTING FROM および ENDING AT 節は、ユーザー指定の値の範囲を基準にしてレコードの集合の読み取りを制限するために使用します。</p> <p>STARTING FROM 節 (=、EQ、EQUAL TO、または [STARTING] FROM) は、読み取り操作の開始値を決めます。開始値を指定すると、指定した値から読み取りが始まります。ファイルに開始値が存在しない場合、次に高い (または DESCENDING 読み取りでは低い) 値が使用されます。より高い (または DESCENDING ではより低い) 値が存在しなければ、ループに入りません。</p> <p>レコードを終了値に制限するには、包括的な範囲を示す用語 THRU、ENDING AT、または TO を持つ ENDING AT 節を指定できます。読み取りディスクリプタフィールドが終了値を超えると、自動的なループ終了が実行されます。TO、THRU および ENDING AT キーワードの基本機能は、非常によく似ていますが、内部的に動作が異なります。</p>
<p><b>THRU/ENDING AT</b></p>	<p>THRU または ENDING AT を使用した場合、開始値だけがデータベースに供給され、データベースによってレコードが返された後に、Natural ランタイムシステムによって終了値のチェックが実行されます。読み取り順が ASCENDING の場合、開始値は READ ループで最初に返される値 (およびレコード) を表すため、開始値としてより低い値を供給し、終了値としてより高い値を供給する必要があります。ただし、逆順読み取り (DESCENDING) を</p>

	<p>呼び出す場合、より高い値は開始値に、そして終了値にはより低い値が出現する必要があります。</p> <p>内部的に、読み取る範囲の終了を決定するために、Natural は終了値を超えるレコードを1件読み取ります。終了値に到達したために READ ループを終了したとき、この最後のレコードは実際には要求した範囲内の最後のレコードではなく、その範囲を超えた最初のレコードである（最後の結果レコードを超えるレコードがファイルに含まれない場合を除く）ことに注意してください。</p> <p>THRU および ENDING AT 節は、READ または HISTOGRAM ステートメントをサポートするすべてのデータベースに対して使用できます。</p>
TO	<p>キーワード TO を使用すると、開始値と終了値の両方がデータベースに送られ、Natural は値の範囲をチェックしません。終了値を超えると、データベースは「ファイルの終わり」に到達した場合と同様に対処し、データベースループを終了します。データベースにより完全な範囲チェックが行われるので、ASCENDING または DESCENDING のいずれの順序でブラウズしていても、常に範囲の小さい方の値が開始値になり、大きい方の値が終了値になります。</p> <p>TO オプションは、基準データベースが Adabas バージョン 7（またはそれ以降）、DB2、VSAM、または DL1 の場合にのみ適用できます。</p>

### THRU/ENDING AT と TO の機能上の違いに関する注意

次のリストは、THRU/ENDING AT および TO オプションの機能上の違いを示しています。

THRU/ENDING AT	TO
終了値に達したために READ ループが終了するとき、ビューには「範囲外」の最初のレコードが含まれます。	終了値に達したために READ ループが終了するとき、ビューには指定した範囲の最後のレコードが含まれます。
終了値変数が READ ループ中に修正された場合、次のレコードの読み取り時に新しい値が終了値チェックに使用されます。	終了値変数は READ ループ開始時にのみ評価されます。READ ループ中の以降の修正はすべて無効です。
範囲が不正な場合（例えば、READ .. = 'B' THRU 'A'）、レコードが返されないだけで、データベースエラーは発生しません。	値範囲を降順で供給してはならないため、不正な範囲はデータベースエラー（例：Adabas RC=61）となります。
READ .. DESCENDING が、開始値および終了値と一緒に使用される場合、開始値はファイルでの位置決めに使用され、終了値は Natural で「範囲終了」をチェックするために使用されます。したがって、開始値は終了値以上の値です。	両方の値がデータベースに渡されるため、それらは昇順で出現する必要があります。つまり、昇順または降順で読み取っているかどうかに関係なく、開始値は終了値以下の値です。
範囲のオーバーフローをチェックするために、ディスクリプタ値は基準データベースビューに出現する必要があります。つまり、それはレコードバッファに返される必要があります。	ディスクリプタは、返されるレコードフィールドに必要ではありません。
Adabas マルチバリューフィールド（MU フィールド）またはサブ/スーパー/ハイパーディスクリプタに対し	MU フィールドおよびサブ/スーパー/ハイパーディスクリプタに対して終了値を指定できます。

THRU/ENDING AT	TO
て終了値チェックはできません。また、プログラムコンパイル時に構文エラー NAT0160 を引き起こします。	
すべてのデータベースに使用できます。	Adabasバージョン7（またはそれ以降）、DB2、VSAM、またはDL/Iに対してのみ使用できます。

## WHERE 節

WHERE *logical-condition*

WHERE 節は、追加の選択条件 (*logical-condition*) を指定するために使用できます。この条件は、値が読み取られた後、値に対する処理が実行される前に評価されます (AT BREAK 評価を含む)。

*logical-condition* の構文については、『プログラミングガイド』の「論理条件基準」で説明しています。

WHERE 節を含む READ ステートメントに LIMIT ステートメントや処理回数の制限を指定した場合、WHERE 節で排除されたレコードは制限数にカウントされません。

## READ で使用可能なシステム変数

Natural システム変数 \*ISN および \*COUNTER が READ ステートメントで使用可能です。

これらのシステム変数は、P10 のフォーマットと長さで定義されています。このフォーマット／長さは変更できません。

システム変数の使用方法について次に説明します。

<b>*ISN</b>	<p>システム変数 *ISN には、現在処理中のレコードの Adabas ISN が入ります。</p> <p><b>注意:</b></p> <ol style="list-style-type: none"> <li>VSAM データベースでは、現在処理中のレコードの RRN (RRDSデータセット) または RBA (ESDSデータセット) が *ISN に入ります。</li> <li>Tamino では、XML オブジェクト ID が *ISN に入ります。</li> <li>DL/I、SQL データベースおよび Entire System Server では、*ISN は使用できません。</li> </ol>
<b>*COUNTER</b>	システム変数 *COUNTER には、入力した処理ループの回数が含まれます。

## 例

- 例 1 - READ ステートメント
- 例 2 - READ WITH REPOSITION
- 例 3 - READ および FIND ステートメントの結合
- 例 4 - DESCENDING オプション
- 例 5 - VARIABLE オプション
- 例 6 - DYNAMIC オプション
- 例 7 - STARTING WITH ISN 節

### 例 1 - READ ステートメント

```

** Example 'REAEX1S': READ (structured mode)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
1 VEHIC-VIEW VIEW OF VEHICLES
  2 PERSONNEL-ID
  2 MAKE
END-DEFINE
*
LIMIT 3
*
WRITE 'READ IN PHYSICAL SEQUENCE'
READ EMPLOY-VIEW IN PHYSICAL SEQUENCE
  DISPLAY NOTITLE PERSONNEL-ID NAME *ISN *COUNTER
END-READ
*
WRITE / 'READ IN ISN SEQUENCE'
READ EMPLOY-VIEW BY ISN STARTING FROM 1 ENDING AT 3
  DISPLAY          PERSONNEL-ID NAME *ISN *COUNTER
END-READ
*
WRITE / 'READ IN NAME SEQUENCE'
READ EMPLOY-VIEW BY NAME
  DISPLAY          PERSONNEL-ID NAME *ISN *COUNTER
END-READ
*
WRITE / 'READ IN NAME SEQUENCE STARTING FROM ''M'' '
READ EMPLOY-VIEW BY NAME STARTING FROM 'M'
  DISPLAY          PERSONNEL-ID NAME *ISN *COUNTER
END-READ
*
END

```

プログラム **REAEX1S** の出力：

PERSONNEL ID	NAME	ISN	CNT
-----			
READ IN PHYSICAL SEQUENCE			
50005800	ADAM	1	1
50005600	MORENO	2	2
50005500	BLOND	3	3
READ IN ISN SEQUENCE			
50005800	ADAM	1	1
50005600	MORENO	2	2
50005500	BLOND	3	3
READ IN NAME SEQUENCE			
60008339	ABELLAN	478	1
30000231	ACHIESON	878	2
50005800	ADAM	1	3
READ IN NAME SEQUENCE STARTING FROM 'M'			
30008125	MACDONALD	923	1
20028700	MACKARNESS	765	2
40000045	MADSEN	508	3

レポートモードの例はライブラリ SYSEXRM のプログラム **REAEX1R** を参照してください。

## 例 2 - READ WITH REPOSITION

```

DEFINE DATA LOCAL
1 MYVIEW VIEW OF ...
  2 NAME
1 #STARTVAL (A20) INIT <'A'>
1 #ATTR (C)
END-DEFINE
...
SET KEY PF3
...
READ MYVIEW WITH REPOSITION BY NAME = #STARTVAL
INPUT (IP=OFF AD=0) 'NAME:' NAME /
  'Enter new start value for repositioning:' #STARTVAL (AD=MT CV=#ATTR) /
  'Press PF3 to stop'
IF *PF-KEY = 'PF3'
  THEN STOP
END-IF
IF #ATTR MODIFIED
  THEN ESCAPE TOP REPOSITION
END-IF

```



```
END-READ
```

```
...
```

```
DEFINE DATA LOCAL
```

```
1 MYVIEW VIEW OF ...
```

```
2 NAME
```

```
1 #STARTVAL (A20) INIT <'A'>
```

```
1 #ATTR (C)
```

```
END-DEFINE
```

```
...
```

```
SET KEY PF3
```

```
...
```

```
READ MYVIEW WITH REPOSITION BY NAME = #STARTVAL
```

```
INPUT (IP=OFF AD=0) 'NAME:' NAME /
```

```
'Enter new start value for repositioning:' #STARTVAL (AD=MT CV=#ATTR) /
```

```
'Press PF3 to stop'
```

```
IF *PF-KEY = 'PF3'
```

```
THEN STOP
```

```
END-IF
```

```
IF #ATTR MODIFIED
```

```
THEN RESET *COUNTER
```

```
END-IF
```

```
END-READ
```

```
...
```

### 例 3 - READ および FIND ステートメントの結合

次の例では、まず EMPLOYEES ファイルから、ディスクリプタ NAME の値で論理順にレコードを読み取ります。次に、EMPLOYEES ファイルの PERSONNEL-ID (人事番号) を検索条件として、VEHICLES ファイルを FIND ステートメントで検索します。結果のレポートには、読み取られた従業員の名前 (EMPLOYEES ファイルから読み取られる) と、この従業員が所有する自動車のモデル (VEHICLES ファイルから読み取られる) が示されます。1 人で何台も自動車を所有している場合、同一名で複数行が出力されます。

```
** Example 'REAEX2': READ and FIND combination
```

```
*****
```

```
DEFINE DATA LOCAL
```

```
1 EMPLOY-VIEW VIEW OF EMPLOYEES
```

```
2 PERSONNEL-ID
```

```
2 FIRST-NAME
```

```
2 NAME
```

```
2 CITY
```

```
1 VEH-VIEW VIEW OF VEHICLES
```

```
2 PERSONNEL-ID
```

```
2 MAKE
```

```
END-DEFINE
```

```
*
```

```
LIMIT 10
```

```
*
```

```
RD. READ EMPLOY-VIEW BY NAME STARTING FROM 'JONES'
```

```

SUSPEND IDENTICAL SUPPRESS
FD. FIND VEH-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (RD.)
  IF NO RECORDS FOUND
    ENTER
  END-NOREC
  DISPLAY NOTITLE (ES=OFF IS=ON ZP=ON AL=15)
    PERSONNEL-ID (RD.)
    FIRST-NAME (RD.)
    MAKE (FD.) (IS=OFF)

  END-FIND
END-READ
END

```

プログラム **REAEX2** の出力：

PERSONNEL ID	FIRST-NAME	MAKE
20007500	VIRGINIA	CHRYSLER
20008400	MARSHA	CHRYSLER
		CHRYSLER
20021100	ROBERT	GENERAL MOTORS
20000800	LILLY	FORD
		MG
20001100	EDWARD	GENERAL MOTORS
20002000	MARTHA	GENERAL MOTORS
20003400	LAUREL	GENERAL MOTORS
30034045	KEVIN	DATSUN
30034233	GREGORY	FORD
11400319	MANFRED	

#### 例 4 - DESCENDING オプション

```

** Example 'READSCND': READ (with DESCENDING SEQUENCE)
*****
DEFINE DATA LOCAL
1 EMPL VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 BIRTH
END-DEFINE
*
READ (10) EMPL IN DESCENDING SEQUENCE BY NAME FROM 'ZZZ'
  DISPLAY *ISN NAME FIRST-NAME BIRTH (EM=YYYY-MM-DD)
END-READ
END

```

## 例 5 - VARIABLE オプション

```

** Example 'REAVSEQ': READ (with VARIABLE SEQUENCE)
*****
DEFINE DATA LOCAL
1 EMPL VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 BIRTH
*
1 #DIR          (A1)
1 #STARTVALUE  (A20)
END-DEFINE
*
SET KEY PF7 PF8
*
INPUT 'Select READ direction'
  // 'Press' 08T 'PF7' (I)                21T 'to read backward'
  /         08T 'PF8' (I) 'or' 'ENTER' (I) 21T 'to read forward'
*
IF *PF-KEY = 'PF7'
  MOVE 'D' TO #DIR
  MOVE 'ZZZ' TO #STARTVALUE
ELSE
  MOVE 'A' TO #DIR
  MOVE 'A' TO #STARTVALUE
END-IF
*
READ (10) EMPL IN VARIABLE #DIR SEQUENCE
          BY NAME FROM #STARTVALUE
  DISPLAY *ISN NAME FIRST-NAME BIRTH (EM=YYYY-MM-DD)
END-READ
END

```

## 例 6 - DYNAMIC オプション

```

DEFINE DATA LOCAL
1 #DIRECTION (A1) INIT <'A'> /* 'A' = ASCENDING
1 #EMPVIEW VIEW OF EMPLOYEES
2 NAME
...
END-DEFINE
...
READ #EMPVIEW IN DYNAMIC #DIRECTION SEQUENCE BY NAME = 'SMITH'
  INPUT (AD=0) NAME
    / 'Press PF7 to scroll in DESCENDING sequence'
    / 'Press PF8 to scroll in ASCENDING sequence'
  ..
  IF *PF-KEY = 'PF7' THEN MOVE 'D' TO #DIRECTION END-IF
  IF *PF-KEY = 'PF8' THEN MOVE 'A' TO #DIRECTION END-IF

```

END-READ

...

**例 7 - STARTING WITH ISN 節**

```

** Example 'REASISND': READ (with STARTING WITH ISN)
*****
DEFINE DATA LOCAL
1 EMPL VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 BIRTH
*
1 #DIR      (A1)
1 #STARTVAL (A20)
1 #STARTISN (N8)
END-DEFINE
*
SET KEY PF3 PF7 PF8
*
MOVE 'ADKINSON' TO #STARTVAL
*
READ (9) EMPL BY NAME = #STARTVAL
  WRITE *ISN NAME FIRST-NAME BIRTH (EM=YYYY-MM-DD) *COUNTER
  IF *COUNTER = 5 THEN
    MOVE NAME TO #STARTVAL
    MOVE *ISN TO #STARTISN
  END-IF
END-READ
*
#DIR := 'A'
*
REPEAT
  READ EMPL IN VARIABLE #DIR BY NAME = #STARTVAL
    STARTING WITH ISN = #STARTISN
  MOVE NAME TO #STARTVAL
  MOVE *ISN TO #STARTISN
  INPUT NO ERASE (IP=OFF AD=0)
  15/01 *ISN NAME FIRST-NAME BIRTH (EM=YYYY-MM-DD)
  // 'Direction:' #DIR
  // 'Press PF3 to stop'
  / ' PF7 to go step back'
  / ' PF8 to go step forward'
  / ' ENTER to continue in that direction'
/*
IF *PF-KEY = 'PF7' AND #DIR = 'A'
  MOVE 'D' TO #DIR
  ESCAPE BOTTOM
END-IF
IF *PF-KEY = 'PF8' AND #DIR = 'D'
  MOVE 'A' TO #DIR

```

```
    ESCAPE BOTTOM
  END-IF
  IF *PF-KEY = 'PF3'
    STOP
  END-IF
END-READ
/*
IF *COUNTER(0290) = 0
  STOP
END-IF
END-REPEAT
END
```

---

# 101

## READ WORK FILE

---

▪ 機能 .....	694
▪ 構文説明 .....	695
▪ フィールド長 .....	698
▪ ラージおよびダイナミック変数の処理 .....	698
▪ 例 .....	699

ストラクチャードモード構文

```

READ [WORK FILE] work-file-number [ONCE]
  {
    RECORD operand1
    [AND] [SELECT] { [ OFFSET n ] ... operand2 } ... }
  [GIVING LENGTH operand3]
  [ AT [END] [OF] [FILE] ]
  [ statement ... ]
  END-ENDFILE
  statement ...
END -WORK
    
```

レポートモード構文

```

READ [WORK FILE] work-file-number [ONCE]
  {
    RECORD { operand1 [FILLER nX] } ...
    [AND] [SELECT] { [ OFFSET n ] ... operand2 } ... }
  [GIVING LENGTH operand3]
  [ AT [END] [OF] [FILE] ] { statement
  DO statement ... DOEND }
  statement ...
[LOOP]
    
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CLOSE WORK FILE](#) | [DEFINE WORK FILE](#) | [WRITE WORK FILE](#)

関連機能グループ：「[ワークファイル/PC ファイルの制御](#)」

## 機能


READ WORK FILE ステートメントは、Adabas 以外の物理順ワークファイルからデータを読み取るために使用します。データはワークファイルから順次読み取られます。データがどのように読み取られるかは、そのデータがワークファイルにどのように書き込まれたということとは無関係です。

このステートメントは、Com-plete、CICS、CMS、TSO、TIAM、またはバッチモード環境で実行されるプログラムでのみ使用できます。適切な JCL またはシステムコマンドを実行してワー



クファイルを割り当てる必要があります。詳細については、『オペレーション』ドキュメントを参照してください。ワークファイル割り当ての詳細については、『パラメータリファレンス』ドキュメントでプロファイルパラメータ `WORK` を参照してください。

READ WORK FILE は、ワークファイルの全レコードを読み取るための処理ループを開始し、実行します。READ WORK FILE ループでは自動ブレイク処理を行うことができます。

 **注意:**

1. READ WORK FILE ステートメントの実行中にエンドオブファイル条件になると、Natural はワークファイルを自動的にクローズします。
2. Entire Connection の場合：Entire Connection ワークファイルを読み取る場合、READ WORK FILE 処理ループ内に I/O ステートメントを指定しないでください。
3. Unicode およびコードページのサポートについては、『Unicode およびコードページのサポート』ドキュメントの「メインフレームプラットフォーム上のワークファイルと出力ファイル」を参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A G	A U N P I F B D T L C G	可	可
<i>operand2</i>	S A G	A U N P I F B D T L C	可	可
<i>operand3</i>	S	I	可	可

フォーマット C は Natural Connection では無効です。

「[フィールド長](#)」も参照してください。

構文要素の説明：

<i>work-file-number</i>	<p>ワークファイル番号：</p> <p>(Natural に定義した) 読み取るワークファイルの番号。</p> <p>変数インデックス範囲：</p>
-------------------------	---

	<p>ワークファイルから配列を読み取る場合、配列に変数インデックス範囲を指定できません。次に例を示します。</p> <pre>READ WORK FILE work-file-number #ARRAY (I:J)</pre>
<p><b>ONCE</b></p>	<p><b>ONCE</b> オプション：</p> <p>ONCE は、1 レコードのみを読み取るために使用します。処理ループは開始されません（したがってループを閉じるキーワード END-WORK または LOOP は指定できません）。ONCE を指定する場合、AT END OF FILE 節も使用する必要があります。</p> <p>ONCE 指定のある READ WORK FILE ステートメントがユーザー開始の処理ループで制御される場合、処理ループの終了前にワークファイルにエンドオブファイル条件が検出されることもあります。ワークファイルから読み取られたすべてのフィールドには、最後に読み取られたレコードの値が残っています。その後、ワークファイルは、READ WORK FILE ONCE の次回実行時に読み取られる最初のレコードに再位置決めされます。</p>
<p><b>RECORD</b> <i>operand1</i> <b>FILLER</b> <i>nX</i></p>	<p><b>RECORD</b> オプション：</p> <p>RECORD を指定した場合、読み取られた各レコードの全フィールドが以降の処理で使用可能になります。レコードのレイアウトに対応するオペランドリスト (<i>operand1</i>) を提供する必要があります。</p> <p>FILLER <i>nX</i> の指定があると、入力データ内で <i>n</i> バイト分がスキップされます。RECORD 節で定義されたレコードは、連続ストレージ内にある必要があります。FILLER はストラクチャードモードでは使用できません。</p> <p>ストラクチャードモードまたは DEFINE DATA ステートメントで定義したレコードを使用する場合、1つのフィールド（またはグループ）だけが使用できます。この場合、FILLER は指定できません。</p> <p>レコードに含まれているデータが Natural によってチェックおよび変換されることはありません。数値フィールド内に数値以外のデータが含まれていると、プログラムの異常終了を起こすので、ユーザーは正しくレコードレイアウトを記述する必要があります。Natural によるチェックは行われなため、このオプションはシーケンシャルファイルのレコードを処理するには最も速い方法です。レコードが読み取られる前は、<i>operand1</i> で定義したレコードエリアは空白で埋められます。したがって、エンドオブファイル条件では空白のエリアが返されます。短いレコードには空白が付加されます。</p> <p>次の場合、<b>RECORD</b> オプションは使用できません。</p> <ul style="list-style-type: none"> <li>■ Entire Connection ワークファイルを読み取る場合</li> <li>■ ダイナミック変数を使用する場合</li> </ul> <p>ワークファイルタイプ CSV を使用する場合、RECORD オプションは無視され、処理は SELECT モードに切り替わります。</p>
<p><b>SELECT</b></p>	<p><b>SELECT</b> オプション（デフォルト）：</p>

	<p>SELECT を指定した場合、オペランドリスト ( <i>operand2</i> ) に指定されたフィールドのみが使用可能になります。入力レコード内のフィールドの位置を OFFSET または FILLER 指定で示すことができます。</p>				
	<table border="1"> <tr> <td>OFFSET <i>n</i></td> <td>OFFSET 0 は、レコードの先頭バイトを示します。TYPE UNFORMATTED で定義されたワークファイルに OFFSET を指定することはできません。</td> </tr> <tr> <td>FILLER <i>nX</i></td> <td>入力レコード内で <i>n</i> バイト分スキップすることを示します。</td> </tr> </table>	OFFSET <i>n</i>	OFFSET 0 は、レコードの先頭バイトを示します。TYPE UNFORMATTED で定義されたワークファイルに OFFSET を指定することはできません。	FILLER <i>nX</i>	入力レコード内で <i>n</i> バイト分スキップすることを示します。
OFFSET <i>n</i>	OFFSET 0 は、レコードの先頭バイトを示します。TYPE UNFORMATTED で定義されたワークファイルに OFFSET を指定することはできません。				
FILLER <i>nX</i>	入力レコード内で <i>n</i> バイト分スキップすることを示します。				
	<p>Natural は、各フィールドに選択された値を割り当て、レコードから選択された数値フィールドに定義どおり正しい数値データが含まれているかをチェックします。選択されたフィールドのチェックは Natural で行われるため、このオプションを使用すると、シーケンシャルファイルの処理にオーバーヘッドがかかります。</p> <p>レコードが SELECT オプションで指定した全フィールドを満たさない場合、次のことが適用されます。</p> <ul style="list-style-type: none"> <li>■ フィールドの一部だけが満たされる場合、残りの部分は空白またはゼロにリセットされます。</li> <li>■ まったく満たされなかったフィールドは前の内容のままです。</li> </ul> <p>ファイルタイプ CSV を読み取る場合、OFFSET オプションは無視されます。</p>				
<b>GIVING LENGTH</b> <i>operand3</i>	<p>GIVING LENGTH 節は、読み取るレコードの実際の長さを取得するために使用します。長さ (バイト数) は <i>operand3</i> に返されます。</p> <p><i>operand3</i> は、フォーマット / 長さ I4 で定義する必要があります。</p> <p>ワークファイルが TYPE UNFORMATTED として定義されている場合、返される長さはバイトストリームから読み取られるバイト数 (FILLER オペランドでスキップしたバイトを含む) を示します。</p> <p>ワークファイルタイプ CSV で GIVING LENGTH 節を使用した場合は、GIVING LENGTH で指定したオペランドではレコード内のフィールド数 (レコードの長さではない) が返されます。</p>				
<b>AT END OF FILE</b>	<p>AT END OF FILE 節は ONCE オプションとともに使用できます。ONCE オプションを使用する場合、エンドオブファイル条件になったときに行う処理をこの節で指定する必要があります。</p> <p>ONCE オプションを使用しない場合、エンドオブファイル条件は通常の処理ループ終了と同様に扱われます。</p>				
<b>END -WORK</b>	<p>READ WORK FILE ステートメントを終了するには、Natural 予約語 END-WORK を使用する必要があります。</p>				

## フィールド長

オペランド定義テーブル内のフィールド長は次のように決定されます。

フォーマット	長さ
A、B、I、F	入力レコード中のバイト数は、内部長さ定義と同じです。
N	入力レコード中のバイト数は、小数点の前後の内部桁数の合計です。小数点と符号は、入力レコード中のバイト数には入りません。
P、D、T	入力レコード中のバイト数は、小数点の前後の桁数の合計に1（符号用）を加算し、それを2で割って切り上げたものです。
L	1バイトが使用されます。Cフォーマットのフィールドについては、2バイトが使用されます。

フィールド長の例：

フィールド定義	入力レコード
#FIELD1 (A10)	10 バイト
#FIELD2 (B15)	15 バイト
#FIELD3 (N1.3)	4 バイト
#FIELD4 (N0.7)	7 バイト
#FIELD5 (P1.2)	2 バイト
#FIELD6 (P6.0)	4 バイト

『プログラミングガイド』の「ユーザー定義変数のフォーマットおよび長さ」も参照してください。

## ラージおよびダイナミック変数の処理

ワークファイルタイプ	処理
<b>UNFORMATTED</b>	UNFORMATTED ワークファイルからダイナミック変数を読み取ると、（現在の位置から）ファイルの残りすべてが変数に挿入されます。ファイルが1073741824バイトを超過した場合、最大の1073741824バイトが変数に挿入されます。
<b>FORMATTED</b>	FORMATTED ワークファイルからダイナミック変数を読み取ると、変数は現在定義されている長さ（長さ0を含む）で埋められます。ファイルの終わりに到達すると、現在のフィールドの残りは空白で埋められます。後続のフィールドは変更されません。

## 例

```

** Example 'RWFEX1': READ WORK FILE
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
*
1 #RECORD
  2 #PERS-ID (A8)
  2 #NAME (A20)
END-DEFINE
*
FIND EMPLOY-VIEW WITH CITY = 'STUTTGART'
  WRITE WORK FILE 1
    PERSONNEL-ID NAME
END-FIND
*
* ...
*
READ WORK FILE 1 RECORD #RECORD
  DISPLAY NOTITLE #PERS-ID #NAME
END-WORK
*
END

```

プログラム **RWFEX1** の出力：

```

#PERS-ID      #NAME
-----
11100328 BERGHAUS
11100329 BARTHEL
11300313 AECKERLE
11300316 KANTE
11500304 KLUGE
11500308 DIETRICH
11500318 GASSNER
11500343 ROEHM
11600303 BERGER
11600320 BLAETTEL
11500336 JASPER
11100330 BUSH
11500328 EGGERT

```



# 102 REDEFINE

---

▪ 機能 .....	702
▪ 制限事項 .....	702
▪ 構文説明 .....	702
▪ 例 .....	703

```
REDEFINE { operand1 ( { nX
                      { operand2 } ... ) } ...
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

REDEFINE ステートメントは、フィールドを再定義するために使用します。再定義により、1つ以上のユーザー定義変数にできます。

1つの REDEFINE ステートメントで複数のフィールドを再定義できます。

## 制限事項

REDEFINE ステートメントはレポーティングモードでのみ有効です。ストラクチャードモードでフィールドを再定義するには、[DEFINE DATA](#) ステートメントの [REDEFINE](#) 節を使用します。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A G	A U N P I F B D T L C	可	不可
<i>operand2</i>	S A G	A N P I F B D T L C	可	可

構文要素の説明：

<b>REDEFINE</b>	再定義の方法
<i>operand1operand2</i>	<p><i>operand1</i> のバイト位置をフォーマットに関係なく左から右に再定義します。</p> <p><i>operand2</i> のフォーマットには、<i>operand1</i> と異なるフォーマットを指定できます。REDEFINE ステートメントで指定するバイトの位置は、元のフィールド内のデータと合っている必要があります。英数字フィールドが数字フィールドに再定義されて、フォーマット指定に従った数字データが入っていない場合、そのフィールドが使用されるときに異常終了することがあります。</p> <p>再定義したフィールドの再定義</p>



	REDEFINE ステートメントで定義したフィールドを、続いて別の REDEFINE ステートメントで再定義することもできます。
<i>nX</i>	<p>フィルラー指定</p> <p><i>nX</i> 指定は、再定義されるフィールド／変数内のフィルラーバイトを示します。フィールド末尾の <i>nX</i> 指定は任意です。</p>

## 例

- 例 1
- 例 2
- 例 3
- 例 4

### 例 1

ユーザー定義変数 #A (フォーマット／長さ A10) の値は 123ABCDEFGG です。

```
REDEFINE #A (#A1(N3) #A2(A7))
```

#A1 の値は 123 です。 #A2 の値は ABCDEFGG です。

### 例 2

ユーザー定義変数 #B (フォーマット／長さ A10) の値 (16 進形式) は 12345CC1C2C3C4C5C6C7 です。

```
REDEFINE #B (#B1(P4) #B2(A7))
```

#B1 の値は 12345C (16 進形式) です。

#B2 の値は C1C2C3C4C5C6C7 (16 進形式) です。

```
REDEFINE #B (#BB1(B2)8X) or REDEFINE #B(#BB1(B2))
```

#BB1 の値は 1234 (16 進形式) です。




**注意:** パックデータ (P フォーマット) では必要な桁数を指定します。次の式でパック形式のデータのバイト数を求めることができます。

バイト数 = (桁数 + 1) / 2、小数点以下は切り上げ

## 例 3

```
COMPUTE #V (N8.2) = #Y (N10) = ...
REDEFINE #V (3X #A(N3) 2X #P (N2)) #Y (#B(N3) 7X)
```



## 例 4

この例では、YYYYMMDD 形式のシステム変数 \*DATN の値を再定義し、「日/月/年」の順で 3 フィールドに分割して結果を出力しています。

```
MOVE *DATN TO #DATINT (N8)
REDEFINE #DATINT (#YEAR (N4) #MONTH (N2) #DAY (N2))
DISPLAY NOTITLE #DATINT #DAY #MONTH #YEAR
END
```

出力：

```
#DATINT #DAY #MONTH #YEAR
-----
19950108 8 1 1995
```

# 103 REDUCE

---

▪ 機能 .....	706
▪ 構文説明 .....	706

```
REDUCE { dynamic-clause } [GIVING operand5]  
      { array-clause }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[EXPAND](#) | [RESIZE](#)

関連機能グループ：「[ダイナミック変数または X-array のメモリ管理制御](#)」

## 機能

REDUCE ステートメントは、次を縮小するために使用します。

- [ダイナミック変数](#) (*dynamic-clause*) の割り当てられた長さ、または
- [X-array](#) (*array-clause*) のオカレンス数

詳細については、『[プログラミングガイド](#)』の「[ダイナミック変数の使用](#)」、「[X-array](#)」、「[X-Group 配列のストレージ管理](#)」も参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A	A U B	不可	不可
<i>operand2</i>	C S	I	不可	不可
<i>operand3</i>	A G	A N P I F B D T L C G O	可	不可
<i>operand4</i>	C S	U N P I	不可	不可
<i>operand5</i>	S	I4	不可	可

構文要素の説明：

<i>dynamic-clause</i>	REDUCE DYNAMIC VARIABLE ステートメントは、ダイナミック変数 ( <i>operand1</i> ) の割り当てられた長さを、指定した長さ ( <i>operand2</i> ) に縮小します。詳細については、後述の「 <a href="#">Dynamic 節</a> 」を参照してください。
<i>operand1</i>	<i>operand1</i> は、長さが縮小されるダイナミック変数です。
<i>operand2</i>	<i>operand2</i> は、縮小後のダイナミック変数の長さを指定するために使用します。値として、負ではない整数の定数または Integer4 (I4) タイプの変数を指定する必要があります。
<i>array-clause</i>	REDUCE ARRAY ステートメントは、X-array ( <i>operand3</i> ) のオカレンス数を、( <i>dim[,dim[,dim]]</i> ) で指定した上下限に縮小します。詳細については、以下の「 <a href="#">Array 節</a> 」を参照してください。
<i>operand3</i>	<i>operand3</i> は X-array です。X-array のオカレンスを削減できます。配列のインデックス表記はオプションです。各次元でインデックス表記として使用できるのは、全範囲を示す表記 * のみです。
<i>dim operand4</i>	X-array 縮小の上下限表記 ( <i>operand4</i> またはアスタリスク) は、ここで指定します。現在の上下限の値を使用する必要がある場合は、 <i>operand4</i> の代わりにアスタリスク (*) を指定する必要があります。詳細については、後述の「 <a href="#">次元</a> 」を参照してください。
<b>GIVING</b> <i>operand5</i>	GIVING 節を指定しない場合は、エラー発生時に Natural ランタイムエラー処理がトリガされます。  GIVING 節を指定した場合は、 <i>operand5</i> に、エラー発生時は Natural メッセージ番号、成功時はゼロが含まれます。

## Dynamic 節

```
[SIZE OF] DYNAMIC [VARIABLE] operand 1 TO operand2
```

REDUCE DYNAMIC VARIABLE ステートメントは、ダイナミック変数 (*operand1*) の割り当てられた長さを、指定した長さ (*operand2*) に縮小します。指定した長さを超える分のダイナミック変数の割り当て済みメモリは、ステートメントの実行時に即座に解放されます。

ダイナミック変数の現在割り当てられている長さ (\*LENGTH) が指定した長さより長い場合、\*LENGTH は指定した長さに設定され、変数の内容が切り捨てられます (修正は行われません)。指定した長さがダイナミック変数の現在割り当てられている長さより長い場合、ステートメントは無視されます。

## Array 節

```
[OCCURRENCES OF] ARRAY operand3 TO { 0
                                         (dim [, dim [, dim]]) }
```

REDUCE ARRAY ステートメントは、X-array (*operand3*) のオカレンス数を、TO (*dim* [, *dim* [, *dim*]]) で指定した上下限に縮小します。

REDUCE TO 0 (ゼロ) を指定した場合、X-array の全オカレンスが解放されます。つまり、配列全体が削減されます。

REDUCE ステートメントで使用される上限または下限は、配列に定義された対応する上限または下限と正確に同じである必要があります。

例：

```
DEFINE DATA LOCAL
1 #a(I4/1:*)
1 #g(1:*)
  2 #ga(I4/1:*)

1 #i(i4)
END-DEFINE
...

*/ reducing #a (1:10)

REDUCE ARRAY #a TO (1:10)          /* #a is reduced
REDUCE ARRAY #a TO (*:10)         /* to 10 occurrences.

*/ reducing #ga (1:10,1:20)

REDUCE ARRAY #g TO (1:10)         /* 1st dimension is set to (1:10)
REDUCE ARRAY #ga TO (*:*,1:20)   /* 1st dimension is dependent and
                                  /* therefore kept with (*:*)
                                  /* 2nd dimension is set to (1:20)

REDUCE ARRAY #a TO (5:10)         /* This is rejected because the lower index
                                  /* must be 1 or *
REDUCE ARRAY #a TO (#i:10)       /* This is rejected because the lower index
                                  /* must be 1 or *

REDUCE ARRAY #ga TO (1:10,1:20)  /* (1:10) for the 1st dimension is rejected
                                  /* because the dimension is dependent and
                                  /* must be specified with (*:*)
```

詳細については、次を参照してください。

- X-array のストレージ管理
- X-Group 配列のストレージ管理

## 次元

$$\left\{ \begin{array}{c} * \\ \left\{ \begin{array}{c} * \\ operand4 \end{array} \right\} : \left\{ \begin{array}{c} * \\ operand4 \end{array} \right\} \end{array} \right\}$$

X-array 縮小の上下限表記 (*operand4* またはアスタリスク) は、ここで指定します。現在の上下限の値を使用する必要がある場合は、*operand4* の代わりにアスタリスク (\*) を指定できます。\*: \* の代わりに単一のアスタリスクを指定できます。

次元数 (*dim*) は、X-array (1、2、または 3) と正確に一致している必要があります。

REDUCE ステートメントを使用するときは、オカレンス数を減らすことだけが可能です。要求した数が現在割り当てられているオカレンス数よりも大きい場合は無視されます。





# 104 REINPUT

---

■ 機能 .....	712
■ 構文説明 .....	713
■ 例 .....	719

REINPUT [FULL] [( <i>statement-parameters</i> )] { USING HELP WITH-TEXT-option }  [ <i>MARK-option</i> ] [ <i>ALARM-option</i> ]
--

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[DEFINE WINDOW](#) | [INPUT](#) | [SET WINDOW](#)

関連機能グループ：「[対話型処理用の画面生成](#)」

## 機能

REINPUT ステートメントは、[INPUT](#) ステートメントに戻って再実行するために使用します。これは通常、前の [INPUT](#) ステートメントの結果、データ入力が無効であったことを示すメッセージを表示するために使用します。「[例1](#)」を参照してください。

[INPUT](#) ステートメントとそれに対応する REINPUT ステートメントの間に、[WRITE](#) または [DISPLAY](#) ステートメントを実行することはできません。REINPUT ステートメントをバッチモードで使用することはできません。

REINPUT ステートメントが実行されると、[INPUT](#) ステートメントが実行されたときに存在していたサブルーチン、特殊条件、およびループ処理を考慮してプログラムステータスが再位置づけされます（[INPUT](#) ステートメントのステータスがまだ有効な場合）。[INPUT](#) ステートメントの実行後にループが開始され、REINPUT ステートメントがそのループ内にある場合、ループは中断され、REINPUT の結果として再処理された [INPUT](#) ステートメントの後で再開されます。

[INPUT](#) ステートメントの実行後にサブルーチン階層が呼び出され、サブルーチン内で REINPUT が実行された場合、Natural は自動的に全サブルーチンをさかのぼり、[INPUT](#) ステートメントの位置にプログラムステータスを再位置づけします。

[INPUT](#) ステートメントが実行されたステータスがすでに終了してしまった時点では、[INPUT](#) ステートメントをループ、サブルーチン、または特殊条件ブロック内に位置づけ、REINPUT ステートメントを実行することはできません。このエラー条件が検出されると、エラーメッセージが生成され、プログラムの実行が終了します。

「[ダイアログ設計](#)」の「[ステートメント REINPUT/REINPUTFULL](#)（『[プログラミングガイド](#)』）」も参照してください。

注：

入出力フィールド（オプション (AD=M)）が [INPUT](#) ステートメントによって表示されているとき、画面に表示されているデータは、フィールドが「変更された」とみなされる場合にのみ変数に

戻されます。フィールドのステータスは、次の条件のいずれかが当てはまる場合、MODIFIED になります。

- フィールドの内容が変更された（フィールドに異なるデータが入力された）場合。
- 空のフィールドでキー EEOF（フィールド末尾まで削除）が押された場合。
- 空のフィールド、またはフィールド内の最後の非空白文字の後ろに空白が入力された場合。
- プロファイルパラメータ CVMIN が ON に設定されており、フィールドのデータが編集オペレーションによって操作されたが最後には内容が復元された場合（例えば、最初の文字を同じ文字で上書きした場合など）。

最終的に変更されなかったフィールドの内容は、画面から変数に転送されません。

FULL オプションを指定することなく REINPUT ステートメントを実行した場合、入出力フィールドの MODIFIED 状態は影響を受けません。上記のオペレーションのいずれかによって INPUT ステートメントを介してフィールドが操作された場合以外は、フィールドは変更されていないとみなされます。逆に言うと、前述のオペレーションの少なくとも1つが実行された場合は、FULL オプションなしで REINPUT ステートメントにより INPUT ステートメントが何度送られたかにかかわらず、フィールドは変更されたものとして扱われます。

つまり、FULL オプションなしで REINPUT ステートメントによりトリガされた INPUT ステートメントを使用して表示されたフィールド値は、前述のフィールドオペレーションによってフィールドが変更された場合にのみ、変数に転送されます。

属性制御変数（オプション CV）が、INPUT ステートメントの後の MODIFIED でチェックされているフィールドに割り当てられている場合は、プログラムコード内で MODIFIED ステータスをチェックできます。

## 構文説明

<b>REINPUT FULL</b>	<p>REINPUT ステートメントで FULL オプションを使用すると、対応する INPUT ステートメントが完全に再実行されます。</p> <ul style="list-style-type: none"> <li>■ FULL オプションを使用しない通常の REINPUT ステートメントでは、INPUT ステートメントと REINPUT ステートメントの間で変更された変数の内容は表示されません。つまり、画面上のすべての変数は、INPUT ステートメントが初めて実行されたときに保持していた内容を示します。</li> <li>■ REINPUT FULL ステートメントでは、INPUT ステートメントの最初の実行後に行われたすべての変更が、INPUT ステートメントの再実行時に INPUT ステートメント適用されます。つまり、画面上のすべての変数は、REINPUT ステートメントの実行時に保持していた値を含みます。</li> </ul> <p><b>注意:</b> REINPUT FULL によって、入力専用フィールド（AD=A）の内容は再度削除されます。</p>
---------------------	---

	<p>REINPUT FULL ステートメントの別の特徴は、属性制御変数のステータスが NOT MODIFIED にリセットされることです。この処理は普通の REINPUT ステートメントでは行われません。属性制御変数にステータス MODIFIED が割り当てられているかどうかを確認するには、MODIFIED オプションを使用します。</p> <p>「例3 - MARK POSITION を指定した REINPUT FULL」も参照してください。</p>									
statement-parameters	<p>REINPUT ステートメントで指定されるパラメータは、ステートメントに指定した全フィールドに適用されます。</p> <p>フィールドレベルで指定したパラメータ (<b>MARK オプション</b>を参照) は、ステートメントレベルの対応する各パラメータを上書きします。</p> <table border="1"> <thead> <tr> <th colspan="2">REINPUT ステートメントで指定可能なパラメータ</th> <th>指定 (S=ステートメントレベル、E=要素レベル)</th> </tr> </thead> <tbody> <tr> <td>AD</td> <td>属性定義*</td> <td>SE</td> </tr> <tr> <td>CD</td> <td>カラー定義</td> <td>S</td> </tr> </tbody> </table> <p>*AD=Pをステートメントパラメータとして指定した場合、すべてのフィールド (<b>MARK オプション</b>で使用されるフィールドを除く) が保護されます。</p> <p>各セッションパラメータの詳細については、『パラメータリファレンス』を参照してください。</p>	REINPUT ステートメントで指定可能なパラメータ		指定 (S=ステートメントレベル、E=要素レベル)	AD	属性定義*	SE	CD	カラー定義	S
REINPUT ステートメントで指定可能なパラメータ		指定 (S=ステートメントレベル、E=要素レベル)								
AD	属性定義*	SE								
CD	カラー定義	S								
USING HELP	<p>USING HELP 節により、INPUT マップに定義されているヘルプルーチンが呼び出されます。</p> <p>USING HELP を <b>MARK</b> オプション付きで使用すると、MARK オプションに指定された最初のフィールドに対して定義されているヘルプルーチンが呼び出されます。フィールドにヘルプルーチンが定義されていない場合は、マップ用のヘルプルーチンが呼び出されます。</p> <p>例：</p> <pre>REINPUT USING HELP MARK 3</pre> <p>結果として、INPUT マップの3番目のフィールドに対して定義されているヘルプルーチンが呼び出されます。</p>									
WITH-TEXT-option	<p>WITH TEXT オプションは、メッセージ行に表示されるテキストを提供するために使用します。下記の「<b>WITH TEXT オプション</b>」を参照してください。</p>									
MARK-option	<p>MARK オプションを使用して、特定のフィールドをマークすることができます。つまり、REINPUT ステートメントの実行時にカーソルが配置されるフィールドを指定できます。下記の「<b>MARK オプション</b>」を参照してください。</p>									
ALARM-option	<p>このオプションは、REINPUT ステートメントの実行時にアクティブ化される端末の音声アラーム機能を提供します。下記の「<b>ALARM オプション</b>」を参照してください。</p>									

## WITH TEXT オプション

WITH TEXT は、メッセージ行に表示されるテキストを提供するために使用します。これは通常、画面処理やエラー修正のために実行する必要がある操作を示すメッセージです。

```
[WITH] [TEXT] { * operand1 } [(attributes)] [,operand3]...7
                operand2 }
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	N P I B*	可	不可
<i>operand2</i>	C S	A U	可	不可
<i>operand3</i>	C S	A U N P I F B D T L	可	不可

\* *operand1* のフォーマット B は、4 以下の長さでのみ使用できます。

構文要素の説明：

<i>operand1</i>	<p><b>Natural</b> メッセージファイルからのメッセージテキスト：</p> <p><i>operand1</i> は、Natural メッセージファイルから取得するメッセージテキストの番号を表します。</p> <p>ユーザー定義メッセージまたは Natural システムメッセージのいずれかを取得できます。</p> <ul style="list-style-type: none"> <li>■ 最大4桁の正の値（954 など）を指定すると、ユーザー定義メッセージを取得できます。</li> <li>■ 最大4桁の負の値（-954 など）を指定すると、Natural システムメッセージを取得できます。</li> </ul> <p>「例4 - WITH TEXT オプション」も参照してください。</p> <p>Natural メッセージファイルは、関連ドキュメントで説明しているように、SYSERR ユーティリティで作成およびメンテナンスします。</p>
<i>operand2</i>	<p>メッセージテキスト：</p> <p><i>operand2</i> は、メッセージ行に配置されるメッセージを表します。</p> <p>「例4 - WITH TEXT オプション」も参照してください。</p>
<i>attributes</i>	<p><i>operand1</i> / <i>operand2</i> にさまざまな出力属性を割り当てることができます。これらの属性と構文については、以下の「出力属性」で説明します。</p>
<i>operand3</i>	<p>メッセージテキストのダイナミック置換：</p> <p><i>operand3</i> は、英数字、テキスト定数、または変数名を表します。</p>

提供された値は、*operand1* または *operand2* のいずれかで指定されるメッセージテキストの一部を置き換えるために使用されます。

表記 *:n:* は、*operand3* の内容を参照するためにメッセージテキスト内に指定します。*n* には *operand3* のオカレンス (1~7) を指定します。

「例4 - WITH TEXT オプション」も参照してください。

**注意:** 複数の *operand3* を指定する場合は、それぞれをコンマで区切る必要があります。コンマを小数点文字として (セッションパラメータ DC で定義) 使用し、数値定数を *operand3* として指定している場合は、コンマの前後に空白を挿入して、コンマが小数点文字と解釈されないようにします。あるいは、複数の *operand3* を INPUT 区切り文字 (セッションパラメータ ID で定義) で区切ることができます。ただし、ID=/ (スラッシュ) の場合は使用できません。

先頭のゼロまたは末尾の空白は、メッセージに表示される前にフィールド値から削除されます。

## 出力属性

*attributes* は、テキスト表示に使用される出力属性を示します。可能な属性:

```
{ AD=AD-value ... }
{ CD=CD-value ... } ...
```

指定可能なセッションパラメータ値については、『パラメータリファレンス』ドキュメントの該当するセクションを参照してください。

- 「AD - 属性定義」の「フィールド表現」
- CD - カラー定義



**注意:** コンパイラは、実際には1つの出力フィールドに複数の属性値を受け入れます。例えば、「AD=BDI」と指定できます。ただし、この場合は最後の値のみが適用されます。示した例では、値 I のみが有効になり、出力フィールドは強調表示されます。

## MARK オプション

MARK オプションを使用して、特定のフィールドをマークすることができます。つまり、REINPUT ステートメントの実行時にカーソルが配置されるフィールドを指定できます。また、フィールド内の特定の位置をマークすることもできます。さらに、入力保護フィールドにしたり、表示属性やカラー属性を変更できます。

```
MARK [POSITION operand4 [IN]] [FIELD] { { operand5 } [(attributes)] } ...
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand4</i>	C S	N P I	可	不可
<i>operand5</i>	C S A	N P I	可	不可

構文要素の説明：


<i>operand5</i>	<p>マークするフィールド：</p> <p><b>INPUT</b> ステートメントに指定された AD=A または AD=M のフィールド（非保護フィールド）にはすべて、1 から始まるフィールド順番号が <b>Natural</b> によって割り当てられます。<i>operand5</i> には、カーソルを位置づけるフィールド番号を指定します。</p> <p><i>*fieldname</i> の表記は、フィールド名を参照として用いて、（INPUT ステートメントで使用した）フィールドに位置づけるために使用します。</p> <p>該当する INPUT フィールドが配列の場合、配列の 1 つ以上のオカレンスを参照するために一意のインデックスまたはインデックス範囲を指定できます。</p> <pre>INPUT #ARRAY (A1/1:5) ... REINPUT (AD=P) 'TEXT' MARK *#ARRAY (2:3)</pre> <p><i>operand5</i> も配列の場合、<i>operand5</i> の値は INPUT 配列に対するフィールド番号として使用されます。</p> <pre>RESET #X(N2/1:2) INPUT #ARRAY ... ... REINPUT (AD=P) 'TEXT' MARK #X (1:2)</pre>
-----------------	---

<b>MARK POSITION</b>	MARK POSITION では、フィールド内の特定の位置 ( <i>operand4</i> で指定) にカーソルを置くことができます。  「例3 - MARK POSITION を指定した REINPUT FULL」も参照してください。
<i>operand4</i>	<i>operand4</i> ではカーソル位置を指定します。 <i>operand4</i> に小数桁を含めることはできません。
<i>attributes</i>	下記の「属性割り当て」を参照してください。

属性割り当て：

AD=[P]	B	CD =	BL
	C		GR
	D		NE
	I		PI
	N		RE
	U		TU
	V		YE

入力フィールド (AD=A または AD=M) に対しては、入力保護属性 (AD=P) を指定できます。

 **注意:** 出力専用フィールド (AD=0) として定義されたフィールドに対して入力非保護属性は指定できません。

AD=P をステートメントレベルで指定した場合、MARK オプションで指定されたフィールドを除く全フィールドが入力保護フィールドになります。

さらにフィールドの表示属性およびカラー属性を変更できます。これらの属性の詳細については、『パラメタリファレンス』のセッションパラメータ AD および CD を参照してください。

「例2 - 属性を割り当てた REINPUT」も参照してください。

## ALARM オプション

[AND] [SOUND] ALARM

このオプションは、REINPUT ステートメントの実行時にアクティブ化される端末の音声アラーム機能を提供します。この機能を使用するには、対応するハードウェアが使用可能である必要があります。



## 例

- 例 1 - REINPUT ステートメント
- 例 2 - 属性を割り当てた REINPUT
- 例 3 - MARK POSITION を指定した REINPUT FULL
- 例 4 - WITH TEXT オプション

### 例 1 - REINPUT ステートメント

```

** Example 'REIEX1': REINPUT
*****
DEFINE DATA LOCAL
1 #FUNCTION (A1)
1 #PARM      (A1)
END-DEFINE
*
INPUT #FUNCTION #PARM
*
DECIDE FOR FIRST CONDITION
  WHEN #FUNCTION = 'A' AND #PARM = 'X'
    REINPUT 'Function A with parameter X selected.'
    MARK *#PARM
  WHEN #FUNCTION = 'C' THRU 'D'
    REINPUT 'Function C or D selected.'
  WHEN #FUNCTION = 'X'
    STOP
  WHEN NONE
    REINPUT 'Please enter a valid function.'
    MARK *#FUNCTION
END-DECIDE
*
END

```

プログラム **REIEX1** の出力：

```
#FUNCTION A #PARM Y
```

**Enter** キーを押した後：

```
PLEASE ENTER A VALID FUNCTION
#FUNCTION A #PARM Y
```

## 例 2 - 属性を割り当てた REINPUT

```

** Example 'REIEX2': REINPUT (with attributes)
*****
DEFINE DATA LOCAL
1 #A (A20)
1 #B (N7.2)
1 #C (A5)
1 #D (N3)
END-DEFINE
*
INPUT (AD=A) #A #B #C #D
*
IF #A = ' ' OR #B = 0
  REINPUT (AD=P) 'RETYPE VALUES'
          MARK *#A (AD=I CD=RE) /* put cursor on first field
          *#B (AD=U CD=PI) /* and change colours
END-IF
*
END

```

## 例 3 - MARK POSITION を指定した REINPUT FULL

```

** Example 'REIEX3': REINPUT (with FULL and POSITION option)
*****
DEFINE DATA LOCAL
1 #A (A20)
1 #B (N7.2)
1 #C (A5)
1 #D (N3)
END-DEFINE
*
INPUT (AD=M) #A #B #C #D
*
IF #A = ' '
  COMPUTE #B = #B + #D
  RESET #D
END-IF
*
IF #A = SCAN 'TEST' OR = ' '
  REINPUT FULL 'RETYPE VALUES' MARK POSITION 5 IN *#A
END-IF
*
END

```

プログラム **REIEX3** の出力：

```
RETYPE VALUES
```

```
#A          #B          0.00 #C          #D          0
```

#### 例 4 - WITH TEXT オプション

```
** Example 'REIEX4': REINPUT (with TEXT option)
*****
DEFINE DATA LOCAL
01 #NAME   (A8)
01 #TEXT   (A20)
END-DEFINE
*
*
INPUT WITH TEXT 'Enter a program name.' 'Program name:' #NAME
*
IF #NAME = ' '
  REINPUT WITH TEXT 'Input missing. Enter a name.'
END-IF
*
IF #NAME NE MASK (A)
  MOVE 'Invalid input.' TO #TEXT
  REINPUT WITH TEXT ':1: Name must start with a letter.',#TEXT
ELSE
  /* Using Natural error message 7600 for demonstration
  COMPRESS *INIT-USER 'on' *DAT4I INTO #TEXT
  INPUT WITH TEXT *-7600,#NAME,#TEXT 'Input accepted.'
END-IF
END
```



# 105 REJECT

---

このステートメントの詳細については、[ACCEPT/REJECT](#) ステートメントを参照してください。



# 106 RELEASE

---

■ 機能 .....	726
■ 構文説明 .....	726
■ 例 .....	727

```
RELEASE {
  STACK
  SETS [set-name...]
  VARIABLES
}
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[STACK](#) | [FIND with RETAIN option](#) | [DEFINE DATA GLOBAL](#)

## 機能

RELEASE ステートメントは次の目的に使用します。

- Natural スタックのすべてのエントリを解放します。
- [RETAIN](#) 節を含む [FIND](#) ステートメントによって保存された ISN の集合を解放します（Adabas データベースにのみ有効）。
- グローバル変数とアプリケーション独立変数をリセットします。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>set-name</i>	C S	A	不可	不可

構文要素の説明：

<b>RELEASE STACK</b>	現在 Natural スタック内にあるすべてのデータ／コマンドを解放します。
<b>RELEASE SETS</b>	Adabas データベースにのみ適用できます。  <i>set-name</i> を指定せずに RELEASE SETS のみを指定すると、 <a href="#">RETAIN</a> 節を含む <a href="#">FIND</a> ステートメントで保存されたすべての ISN 集合が解放されます。



<b>RELEASE SETS</b>  <i>set-name</i>	特定の単一 ISN 集合を解放します。  <pre>RELEASE SET 'CITY-SET'  MOVE 'CITY-SET' TO #SET(A32) RELEASE SET #SET</pre>
<b>RELEASE VARIABLES</b>	現在のグローバルデータエリアに定義されているすべての変数をそれぞれの初期値にリセットします。また、アプリケーション独立変数 (AIV) はすべて削除されるので、使用できなくなります。

## 例

```
** Example 'RELEX1': FIND (with RETAIN clause and RELEASE statement)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 BIRTH
  2 NAME
*
1 #BIRTH (D)
END-DEFINE
*
MOVE EDITED '19400101' TO #BIRTH (EM=YYYYMMDD)
*
FIND NUMBER EMPLOY-VIEW WITH BIRTH GT #BIRTH
  RETAIN AS 'AGESET1'
IF *NUMBER = 0
  STOP
END-IF
*
FIND EMPLOY-VIEW WITH 'AGESET1' AND CITY = 'NEW YORK'
  DISPLAY NOTITLE NAME CITY BIRTH (EM=YYYY-MM-DD)
END-FIND
*
RELEASE SET 'AGESET1'
*
END
```

プログラム **RELEX1** の出力：

NAME	CITY	DATE	OF BIRTH
-----	-----	-----	-----
RUBIN	NEW YORK	1945-10-27	
WALLACE	NEW YORK	1945-08-04	

# 107 REPEAT

---

▪ 機能 .....	730
▪ 構文説明 .....	730
▪ 例 .....	731

関連ステートメント：[FOR](#) | [ESCAPE](#)

関連機能グループ：「[ループ実行](#)」

## 機能

REPEAT ステートメントは、処理ループを開始するために使用します。

## 構文説明

このステートメントには、2つの異なる構造が可能です。

- **構文 1** を使用すると、ステートメントは 1 回または複数回実行されます。
- **構文 2** を使用すると、ステートメントは 0 回または複数回実行されます。

論理条件の指定位置（ループの最初または最後）によって評価のタイミングが決まります。

論理条件の詳細については、「[論理条件基準](#)」（『[プログラミングガイド](#)』）を参照してください。

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

構文 1：

```
REPEAT
statement ... [ { UNTIL } logical-condition ]
                { WHILE }
END-REPEAT    (ストラクチャードモードのみ)
[ LOOP]       (レポーティングモードのみ)
```

構文 2：

```
REPEAT
[ { UNTIL } logical-condition ] statement...
  { WHILE }
END-REPEAT    (ストラクチャードモードのみ)
[ LOOP]       (レポーティングモードのみ)
```

構文要素の説明：

<b>UNTIL</b>	論理条件が真になるまで、処理ループは継続します。
<b>WHILE</b>	論理条件が真である限り、処理ループは継続します。
<i>logical-condition</i>	論理条件を指定すると、その条件によりいつループ実行を終了するかが決まります。  論理条件を指定しない場合、ループ内で <b>ESCAPE</b> 、 <b>STOP</b> 、または <b>TERMINATE</b> ステートメントを指定し、ループから出る必要があります。
<b>END-REPEAT</b>	REPEAT ステートメントを終了するには、Natural 予約語 <b>END-REPEAT</b> を使用する必要があります。

## 例

- 例 1 - REPEAT
- 例 2 - WHILE および UNTIL オプションの使用

### 例 1 - REPEAT

```

** Example 'RPTX1S': REPEAT (structured mode)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
*
1 #PERS-NR (A8)
END-DEFINE
*
REPEAT
  INPUT 'ENTER A PERSONNEL NUMBER:' #PERS-NR
  IF #PERS-NR = ' '
    ESCAPE BOTTOM
  END-IF
/*
  FIND EMPLOY-VIEW WITH PERSONNEL-ID = #PERS-NR
  IF NO RECORD FOUND
    REINPUT 'NO RECORD FOUND'
  END-NOREC
  DISPLAY NOTITLE NAME
END-FIND
END-REPEAT
*
END

```

## REPEAT

---

プログラム **RPTEX1S** の出力：

```
ENTER A PERSONNEL NUMBER: 11500304
```

人事番号を入力および確定した後：

```
      NAME
-----
KLUGE
```

レポートモードの例はライブラリ SYSEXRМ のプログラム **RPTEX1R** を参照してください。

### 例 2 - WHILE および UNTIL オプションの使用

```
** Example 'RPTEX2S': REPEAT (with WHILE and UNTIL option)
*****
DEFINE DATA LOCAL
1 #X (I1) INIT <0>
1 #Y (I1) INIT <0>
END-DEFINE
*
REPEAT WHILE #X <= 5
  ADD 1 TO #X
  WRITE NOTITLE '=' #X
END-REPEAT
*
SKIP 3
REPEAT
  ADD 1 TO #Y
  WRITE '=' #Y
  UNTIL #Y = 6
END-REPEAT
*
END
```

プログラム **RPTEX2S** の出力：

```
#X:  1
#X:  2
#X:  3
#X:  4
#X:  5
#X:  6
```

```
#Y:  1
#Y:  2
```

---

```
#Y: 3  
#Y: 4  
#Y: 5  
#Y: 6
```

レポートモードの例はライブラリ SYSEXRM のプログラム [RPTEX2R](#) を参照してください。





# 108 REQUEST DOCUMENT

---

■ 機能 .....	736
■ 構文説明 .....	737
■ 受信／送信データのエンコード .....	745
■ 例 .....	747

```

REQUEST DOCUMENT FROM operand1

    WITH
    [USER operand2]
    [PASSWORD operand3]
    [HEADER[[NAME] operand4 [VALUE] operand5]...]

    [
    DATA {
        ALL operand6 [ENCODED [[IN] CODEPAGE operand7]]
        {[NAME] operand8 [VALUE] operand9}...
    }
    ]

    RETURN
    [
    HEADER {
        [ALL operand10]
        [[NAME] operand11 [VALUE] operand12]...
    }
    ]

    [PAGE operand13 [ENCODED [[FOR TYPES[S] operand14...] [IN] CODEPAGE operand15]]]

RESPONSE operand16
[GIVING operand17]

```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[インターネットとXML](#)」

## 機能

REQUEST DOCUMENT ステートメントは、外部システムにアクセスする方法を提供します。『[プログラミングガイド](#)』の「[インターネットおよびXML アクセス用のステートメント](#)」を参照してください。

Unicode サポートの詳細については、『[Unicode およびコードページのサポート](#)』ドキュメントの「[ステートメント](#)」を参照してください。

### プロトコルタイプに対する制限事項

技術的な理由で、HTTPS は z/OS でのみサポートされています。

## クッキーに対する制限事項

HTTPプロトコル下では、クライアントワークステーションの状況についての情報を維持するために、サーバーはクッキーを使用します。

REQUEST DOCUMENT は、インターネットオプション設定を使用して実装されます。これは、セキュリティ設定に依存してクッキーが使用されることを意味しています。

インターネットオプション設定Disabled（使用不可）が設定されている場合、クッキーヘッダー（*operand 4/5*）が送信されても、クッキーは送信されません。

サーバー環境に対して、インターネットオプション設定Promptを使用しないでください。クライアントはプロンプトに答えることができないため、この設定によりサーバーが「ハングアップ」します。

メインフレーム環境ではクッキーはサポートされていないため、無視されます。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A	不可	可
<i>operand2</i>	C S	A	不可	可
<i>operand3</i>	C S	A	不可	可
<i>operand4</i>	C S	A	不可	可
<i>operand5</i>	C S	A N P I F D T L	不可	可
<i>operand6</i>	C S	A U N P I F B D T L	不可	可
<i>operand7</i>	C S	A	不可	可
<i>operand8</i>	C S	A	不可	可
<i>operand9</i>	C S	A N P I F D T L	不可	可
<i>operand10</i>	S	A	不可	可
<i>operand11</i>	C S	A	不可	可
<i>operand12</i>	S	A N P I F B D T L	不可	可
<i>operand13</i>	S	A U B	不可	可
<i>operand14</i>	C S	A	不可	可
<i>operand15</i>	C S	A	不可	可
<i>operand16</i>	S	I4	不可	可
<i>operand17</i>	S	I4	不可	不可

## 構文要素の説明

<b>DOCUMENT</b>	ドキュメントの場所：
<b>FROM</b> <i>operand1</i>	<i>operand1</i> は、ドキュメントにアクセスするための URL です。 <b>注意:</b> 次の情報は、 <i>operand1</i> が http:// または https:// で始まる場合にのみ有効です。
<b>WITH</b>	<b>WITH 節：</b>  この節は、要求に対するオプションのユーザー/パスワード、ヘッダー、およびデータ詳細を指定するために使用します。
<b>USER</b> <i>operand2</i>	ユーザー名：  <i>operand2</i> は、要求に対して使用されるユーザーの名前です。
<b>PASSWORD</b> <i>operand3</i>	ユーザーパスワード：  <i>operand3</i> は、要求に対して使用されるユーザーのパスワードです。
<b>HEADER</b> { <b>[[NAME]</b> <i>operand4</i> <b>[VALUE]</b> <i>operand5</i> }]...	ヘッダー節：  <i>operand4</i> と <i>operand5</i> は、相互に組み合わせた場合にのみ使用できます。 <ul style="list-style-type: none"> <li>■ <i>operand4</i> は、この要求によって送信される HEADER 変数の名前です。</li> <li>■ <i>operand5</i> は、この要求によって送信される HEADER 変数の値です。</li> </ul> <b>注意:</b>  <i>operand4</i> のヘッダー名：  ヘッダー名にキャリッジリターン (CR)、ラインフィード (LF)、またはコロン (:) を含めることはできません。これは、REQUEST DOCUMENT ステートメントではチェックされません。有効なヘッダー名については、HTTP仕様を参照してください。Web インターフェイスとの互換性のために、ヘッダー名はダッシュ (-) でなく下線 ( _ ) で記述できます。(内部的には、下線がダッシュで置換されます。)  <i>operand5</i> のヘッダー値：  ヘッダー値に CR/LF を含めることはできません。これは、REQUEST DOCUMENT ステートメントではチェックされません。有効なヘッダー値とフォーマットについては、HTTP仕様を参照してください。  一般情報  HTTP 要求には、いくつかのヘッダー (例: Request-Method または Content-Type) が必要です。これらのヘッダーは、REQUEST DOCUMENT ステートメントで指定されたパラメータに応じて自動的に生成されます。  「 <a href="#">自動生成ヘッダー</a> 」も参照してください。
<b>DATA</b>	<b>DATA 節：</b>

	特定の DATA 変数名と値（下記の <i>operand8</i> および <i>operand9</i> を参照）、または完全なドキュメント（下記の「 <a href="#">DATA ALL 節</a> 」を参照）を指定できます。								
<b>ALL</b> <i>operand6</i>	<i>operand6</i> は、送信される完全なドキュメントです。この値は、HTTP 要求メソッド PUT で必要です（「 <a href="#">自動生成ヘッダー</a> 」を参照）。  「受信／送信データのエンコード」の「 <a href="#">DATA ALL 節</a> 」を参照してください。								
<b>[ENCODED [[IN] CODEPAGE</b> <i>operand7</i>	<i>operand6</i> は、デフォルトコードページ（システム変数 *CODEPAGE の値）から <i>operand7</i> で指定されたコードページにエンコードされます。  「受信／送信データのエンコード」の「 <a href="#">DATA ALL 節</a> 」を参照してください。								
<b>{[NAME]</b> <i>operand8</i> <b>[VALUE]</b> <i>operand9</i> ...	<b>DATA 変数名と値：</b>  <i>operand8</i> と <i>operand9</i> は、相互に組み合わせた場合にのみ使用できます。  ■ <i>operand8</i> は、この要求によって送信される DATA 変数の名前です。この値は、HTTP 要求メソッド POST で必要です（URL エンコード必須。特にアンパサンド（&）、等号（=）、パーセント記号（%）の文字）。  ■ <i>operand9</i> は、この要求によって送信される DATA 変数の名前です。この値は、HTTP 要求メソッド POST で必要です（URL エンコード必須。特にアンパサンド（&）、等号（=）、パーセント記号（%）の文字）。  制限事項  <i>operand8</i> / <i>operand9</i> が指定され、デフォルトで通信が http:// または https:// の場合、コンテンツタイプ application/x-www-form-urlencoded の要求メソッド POST（「 <a href="#">自動生成メソッド</a> 」を参照）が使用されます。要求中、 <i>operand8</i> / <i>operand9</i> は等号（=）およびアンパサンド（&）の文字で区切られます。したがって、等号（=）およびアンパサンド（&）文字をオペランドに含めることはできません。また、URL エンコードのため、パーセント記号（%）文字を含めることもできません。これらの文字は、「危険」とみなされ、次のようにエンコードする必要があります。  <table border="1"> <thead> <tr> <th>文字</th> <th>URL エンコード構文</th> </tr> </thead> <tbody> <tr> <td>%</td> <td>%25</td> </tr> <tr> <td>&amp;</td> <td>%26</td> </tr> <tr> <td>=</td> <td>%3D</td> </tr> </tbody> </table> 「 <a href="#">URL エンコードに対する一般的な注意</a> 」も参照してください。	文字	URL エンコード構文	%	%25	&	%26	=	%3D
文字	URL エンコード構文								
%	%25								
&	%26								
=	%3D								
<b>RETURN</b>	<b>RETURN 節：</b>  この節は、返される HEADER または PAGE 情報を指定するために使用します。								
<b>HEADER [ALL</b> <i>operand10</i>	<b>RETURN HEADER ALL 節：</b>  この節を指定すると、 <i>operand10</i> に HTTP レスポンスで配信されたすべてのヘッダー値が含まれます。  最初の行にはステータス情報が含まれ、すべての後続行には名前と値のペアとしてヘッダーが含まれます。名前は常にコロン（:）で終了し、値は改行（LF）で終了します。内部的には、すべてのキャリッジリターン（CR）／ラインフィード（LF）は改行（LF）に変換されます。								

<p><b>HEADER</b> [[NAME] <i>operand11</i> <b>[VALUE]</b> <i>operand12</i>...</p>	<p><b>RETURN HEADER NAME/VALUE 節：</b></p> <p>この節を指定すると、特定のヘッダー情報のみが返されます。</p> <p><i>operand11</i> と <i>operand12</i> は、相互に組み合わせた場合にのみ使用できます。</p> <ul style="list-style-type: none"> <li>■ <i>operand11</i> は、この要求で受信された HEADER の名前です。HEADER は HTTP で必要です。</li> <li>■ <i>operand12</i> は、この要求で受信された HEADER の値です。HEADER は HTTP で必要です。</li> </ul> <p><i>operand11</i> に返されるヘッダー名：</p> <p>Web インターフェイスとの互換性のために、ヘッダー名は ダッシュ (-) 文字の代わりに 下線 ( _ ) で記述できます。</p> <p>(内部的には、下線がダッシュで置換されます。) <i>operand11</i> が空白の文字列の場合、ステータス情報が返されます。</p> <pre>HTTP/1.0 200 OK</pre>
<p><b>RETURN PAGE</b></p>	<p><b>RETURN PAGE 節：</b></p> <p>受信データを特定のコードページでエンコードする場合、PAGE 節を使用できます。</p> <p>以下の「受信/送信データのエンコード」の「<a href="#">RETURN PAGE 節</a>」を参照してください。</p>
<p><b>PAGE</b> <i>operand13</i></p>	<p><i>operand13</i> は、この要求に対して返されるドキュメントです。</p> <p>以下の「受信/送信データのエンコード」の「<a href="#">RETURN PAGE 節</a>」を参照してください。</p>
<p><b>[ENCODED</b> [[FOR TYPE[S] <i>operand14</i>...] <b>[IN] CODEPAGE</b> <i>operand15</i>]]</p>	<p><i>operand14</i> は、<i>operand13</i> に返されるドキュメントのエンコードを実行する MIME タイプのリストです。</p> <p>以下の「受信/送信データのエンコード」の「<a href="#">RETURN PAGE 節</a>」を参照してください。</p> <p><i>operand15</i> は、必要に応じて <i>operand13</i> のエンコードに使用されるコードページです。</p> <p><i>operand15</i> の値が空白の場合、<i>operand13</i> は、キーワードサブパラメータ RDCP で定義されたコードページからデフォルトコードページ (A/B) または (U) にエンコードされます。プロファイルパラメータ XML のキーワードサブパラメータ RDCP は、デフォルトの HTML/XML コードページの名前を指定するために使用します。</p> <p>以下の「受信/送信データのエンコード」の「<a href="#">RETURN PAGE 節</a>」を参照してください。</p>
<p><b>RESPONSE</b> <i>operand16</i></p>	<p><b>RESPONSE 節：</b></p> <p>RESPONSE 節は、要求のレスポンスコード番号を表示するために使用します。</p> <p><i>operand16</i> は、要求のレスポンスコード番号です。例えば、200 (要求完了) です。</p> <p>「<a href="#">レスポンス番号の概要 - HTTP/HTTPS 要求</a>」も参照してください。</p>

<b>GIVING</b> <i>operand17</i>	<b>GIVING 節：</b> 要求を実行できなかった場合は、 <i>operand17</i> に Natural エラーが含まれます。
-----------------------------------	---

## 自動生成ヘッダー (operand4/5)

### Request-Method

*operand5* では、HEAD、POST、GET、および PUT がサポートされています。

次の表は、指定されたオペランドに応じた Request-Method の自動的な計算を示しています。

オペランド	Request-Method			
	HEAD	POST	GET	PUT
<b>WITH HEADER</b> ( <i>operand4/operand5</i> )	オプション	オプション	オプション	オプション
<b>WITH DATA</b> ( <i>operand7/operand8</i> )	指定なし	指定	指定なし	オプション ALL ( <i>operand6</i> ) 指定の場合のみ
<b>RETURN HEADER</b> ( <i>operand10~operand12</i> )	指定	オプション	オプション	オプション
<b>RETURN PAGE</b> ( <i>operand13</i> )	指定なし	指定	指定	オプション

### Content-Type

要求メソッドが POST の場合、コンテンツタイプヘッダーを HTTP 要求で配信する必要があります。コンテンツタイプが明示的に設定されていない場合、*operand5* に対して次の値が自動的に生成されます。

```
application/x-www-form-urlencoded
```



**注意:** 自動的に生成されたヘッダーを上書きすることは可能です。Natural はエラーについてそれらをチェックしません。予期しないエラーが起こるかもしれません。

## URL エンコードに対する一般的な注意

コンテンツタイプ `application/x-www-form-urlencoded` で POST データを送信するとき、一定の文字が URL エンコードによって表される必要があります。これは、%16 進文字コードの文字で代用することを意味しています。URL エンコードが必要なときとその理由の詳細については、RFC 1630、RFC 1738、RFC 1808 を参照してください。ここでは、いくつかの基本的な詳細を示します。すべての非 ASCII 文字（つまり、ASCII 文字でもない有効な ISO 8859/1 文字）は URL エンコードする必要があります。例えば、ファイル `köln.html` は URL で `k%F6ln.html` として表示されます。

Web ページが電子メールによって要求されるとき、いくつかの文字が「危険」と考えられます。

これらの文字は次のとおりです。

文字	URL エンコード構文
タブ文字	%09
スペース文字	%20
[	%5B
\	%5C
]	%5D
^	%5E
`	%60
{	%7B
	%7C
}	%7D
~	%7E

URL を記述するとき、これらの文字の URL エンコードが必要です。

いくつかの文字は URL で特別な意味を持っています。例えば、コロン (:) は URL の残りから URL スキームを区別し、2つのスラッシュ (//) は URL が共通インターネットスキーム構文とパーセント記号 (%) に対応することを示します。一般的に、これらの文字がファイル名の一部として出現するときには、それらを URL の特別な意味と区別するために、URL エンコードが必要です。これは単純化のためです。詳細については、RFC を参照してください。



これらの文字は次のとおりです。

文字	URL エンコード構文
"	%22
#	%23
%	%25
&	%26
+	%2B
,	%2C
/	%2F
:	%3A
<	%3C
=	%3D
>	%3E
?	%3F
@	%40

## レスポンス番号の概要 - HTTP/HTTPS 要求

ステータス	値	レスポンス
STATUS CONTINUE	100	要求の継続は OK です
STATUS SWITCH_PROTOCOLS	101	サーバーはアップグレードヘッダーでプロトコルを切り替えました
STATUS OK	200	要求は完了しました
STATUS CREATED	201	オブジェクトが作成されました (理由 = 新しい URL)
STATUS ACCEPTED	202	非同期完了 (TBS)
STATUS PARTIAL	203	部分的な完了
STATUS NO_CONTENT	204	返すための情報がありません
STATUS RESET_CONTENT	205	要求は完了しましたがフォームをクリアしてください
STATUS PARTIAL_CONTENT	206	部分的に GET を実行しました
STATUS AMBIGUOUS	300	サーバーは、何を戻すかを決定できませんでした
STATUS MOVED	301	オブジェクトは恒久的に移動しました
STATUS REDIRECT	302	オブジェクトは一時的に移動しました
STATUS REDIRECT_METHOD	303	新規アクセスメソッドを使用しないリダイレクション
STATUS NOT_MODIFIED	304	If-modified-since は修正されませんでした
STATUS USE_PROXY	305	プロキシへのリダイレクション (使用するプロキシはロケーションヘッダーで指定)
STATUS REDIRECT_KEEP_VERB	307	HTTP/1.1 : 同じ verb を保持してください

ステータス	値	レスポンス
STATUS BAD_REQUEST	400	無効な構文
STATUS DENIED	401	アクセスが拒否されました
STATUS PAYMENT_REQ	402	支払いが必要です
STATUS FORBIDDEN	403	要求が許可されていません
STATUS NOT_FOUND	404	オブジェクトは見つかりませんでした
STATUS BAD_METHOD	405	メソッドは許可されていません
STATUS NONE_ACCEPTABLE	406	クライアントで許容できるレスポンスが見つかりませんでした
STATUS PROXY_AUTH_REQ	407	プロキシ認証が必要です
STATUS REQUEST_TIMEOUT	408	要求待ちサーバーのタイムアウト
STATUS CONFLICT	409	ユーザーはさらに詳細な情報で再度サブミットする必要があります
STATUS GONE	410	リソースが使用できなくなっています
STATUS LENGTH_REQUIRED	411	サーバーは、長さなしの要求の受け入れを拒否しました
STATUS PRECOND_FAILED	412	要求に指定された必須条件が失敗しました
STATUS REQUEST_TOO_LARGE	413	要求エンティティが大きすぎです
STATUS URL_TOO_LONG	414	要求 URL が長すぎです
STATUS UNSUPPORTED_MEDIA	415	未サポートのメディアタイプ
STATUS SERVER_ERROR	500	内部サーバーエラー
STATUS NOT_SUPPORTED	501	「Required」 はサポートされていません
STATUS BAD_GATEWAY	502	ゲートウェイからエラーレスポンスを受信しました
STATUS SERVICE_UNAVAIL	503	一時的に過負荷がかかっています
STATUS GATEWAY_TIMEOUT	504	ゲートウェイ待ちのタイムアウト
STATUS VERSION_NOT_SUP	505	HTTP バージョンがサポートされていません

### レスポンス 301～303 (リダイレクション)

リダイレクションは、要求された URL が移動したことを意味しています。レスポンスとして、LOCATION という名前の Return Header が表示されます。このヘッダーには、要求されたページが移動した URL が含まれます。新しい REQUEST DOCUMENT 要求は、移動したページを検索するために使用できます。

HTTP ブラウザは新しい URL に自動的にリダイレクトしますが、REQUEST DOCUMENT ステートメントはリダイレクションを自動的に処理しません。

### レスポンス 401 (拒否)

レスポンス Access Denied は、有効なユーザー ID とパスワードが要求で提供される場合のみ、要求されたページにアクセスできることを意味しています。レスポンスとして、WWW-AUTHENTICATE という名前の Return Header が、この要求に必要な領域で提供されます。

HTTPブラウザは、通常、ユーザーIDとパスワードによってダイアログを表示しますが、REQUEST DOCUMENT ステートメントではダイアログは表示されません。

## 受信／送信データのエンコード

REQUEST DOCUMENT ステートメントを使用したデータ転送では、通常、コードページ変換は呼び出されません。特定のコードページで受信／送信データをエンコードする場合、DATA ALL 節または RETURN PAGE 節でその指定を行います。

- DATA ALL 節
- RETURN PAGE 節

### DATA ALL 節

送信データのエンコードには、DATA ALL 節を使用します。

```
ALL operand6 [ENCODED [[IN] CODEPAGE operand7]]
```

構文の説明：

ALL operand6	operand6は、送信される完全なドキュメントです。この値は、HTTP 要求メソッド PUT が必要です（「 <a href="#">自動生成ヘッダー</a> 」を参照）。
[ENCODED [[IN] CODEPAGE operand7]]	operand6は、デフォルトコードページ（システム変数 *CODEPAGE の値）から operand7 で指定されたコードページにエンコードされます。

### RETURN PAGE 節

受信データのエンコードには、RETURN PAGE 節を使用します。


```
[PAGE operand13 [ENCODED [[FORTYPE[S] operand14...] [IN] CODEPAGE operand15]]]
```

HTTP/HTTPS 要求のレスポンスとして、受信データにバイナリデータ（例えば image/gif）または文字データ（例えば text/html）が含まれることがあります。REQUEST DOCUMENT ステートメントは、レスポンスとともに、要求したドキュメントのコンテンツタイプ（MIME タイプ）を指定するパラメータを受け取ります。このパラメータには、ドキュメントのエンコードに使用されるコードページに関する情報が含まれることがあります。

この節により、Natural セッションのデフォルトコードページ（システム変数 \*CODEPAGE の値）への自動変換が実行されます。

構文の説明：

<b>RETURN PAGE</b> <i>operand13</i>	返されるページに対してエンコードは行われません。つまり、ページのエンコードは HTTP サーバーから配信されたときの状態のまま変わりません。
<b>RETURN PAGE ENCODED</b> <i>operand13</i>	返される MIME タイプにエンコードが含まれる場合、 <i>operand13</i> はこのコードページからデフォルトコードページ (A/B) または (U) にエンコードされます。下記の注を参照してください。
<b>RETURN PAGE ENCODED [IN] CODEPAGE</b> <i>operand15</i>	返される MIME タイプにエンコードが含まれない場合、 <i>operand13</i> は <i>operand15</i> で定義されたコードページからデフォルトコードページ (システム変数 *CODEPAGE の値) (A/B) または (U) にエンコードされます。
<b>RETURN PAGE [ENCODED [[FOR TYPE[S]</b> <i>operand14...</i> ] [IN] CODEPAGE <i>operand15</i> ]]	返される MIME タイプにエンコードが含まれない場合、その MIME タイプが <i>operand14</i> で指定されたタイプのいずれかと一致しているかどうか、追加のチェックが行われます。一致している場合、 <i>operand13</i> は <i>operand15</i> で定義されたコードページからデフォルトコードページ (A/B) または (U) にエンコードされます。

 **注意:** 「返される MIME タイプにエンコードが含まれる」ということは、HTTPサーバーから charset=節 (例: charset=ISO-8859-1) を含むコンテンツタイプヘッダーが返されることを意味します。

### RETURN PAGE ENCODED 節の使用サンプル

1. サーバーからヘッダー 'Content-type: text/html; charset=UTF-8' が返されます。

プログラムコードサンプル 1:

```
... RETURN PAGE operand13
```

処理結果:

*operand13* は UTF-8 エンコードのまま変わりません。

プログラムコードサンプル 2:

```
... RETURN PAGE operand13 ENCODED [...]
```

処理結果:

*operand13* は、後で指定される *operand15* および *operand14* に関係なく、UTF-8 からデフォルトコードページに変換されます。つまり、返されたコンテンツタイプヘッダーで有効なエンコードが見つかったため、*operand14* と *operand15* は評価されません。

2. サーバーからヘッダー 'Content-type: text/xml' が返されます。

プログラムコードサンプル 1:

```
... RETURN PAGE operand13 ENCODED
```

処理結果：

コンテンツタイプヘッダーに有効なエンコードが含まれていないため、*operand13* は未変換のまま変わりません。

プログラムコードサンプル 2：

```
... RETURN PAGE operand13 ENCODED FOR TYPES 'text/xml' IN CODEPAGE 'USASCII'
```

処理結果：

*operand13* は、USASCII コードページからデフォルトコードページに変換されます。この場合、受け取るページのエンコードに関するプログラムの想定に基づいて変換が行われます。

プログラムコードサンプル 3：

```
... RETURN PAGE operand13 ENCODED FOR TYPES 'text/html' IN CODEPAGE ' '
```

処理結果：

*operand14* で指定された MIME タイプ 'text/html' が、コンテンツタイプヘッダーで返された MIME タイプ 'text/xml' と一致しないため、*operand13* は未変換のまま変わりません。

プログラムコードサンプル 4：

```
... RETURN PAGE operand13 ENCODED IN CODEPAGE ' '
```

処理結果：

*operand13* は、プロファイルパラメータ XML のサブパラメータ RDCP で指定されたデフォルトコードページからデフォルトコードページに変換されます。



**注意:** RDCP サブパラメータのデフォルト値は 'ISO-8859-1' で、明示的な指定が何もない場合に適用されます。『パラメータリファレンス』の「XML - PARSE XML および REQUEST DOCUMENT ステートメントの有効化」も参照してください。

## 例

- 例 1 - 一般的な要求
- 例 2 - 単純な Get 要求 (データなし)
- 例 3 - 単純な Head 要求 (戻りページなし)
- 例 4 - 単純な Post 要求 (デフォルト)

- 例 5 - 単純な Put 要求 (すべてのデータを処理)



注意: このステートメントのダイアログ例 V5-RDOC がライブラリ SYSEXV にあります。

### 例 1 - 一般的な要求

```
REQUEST DOCUMENT FROM "http://bo1sap1:5555/invoke/sap.demo/handle_RFC_XML_POST"  
WITH  
  USER #User PASSWORD #Password  
  DATA  
  NAME 'XMLData'          VALUE #Queryxml  
  NAME 'repServerName' VALUE 'NT2'  
RETURN  
  PAGE #Resultxml  
RESPONSE #rc
```

### 例 2 - 単純な Get 要求 (データなし)

```
REQUEST DOCUMENT FROM "http://pcnatweb:8080"  
RETURN  
  PAGE #Resultxml  
RESPONSE #rc
```

### 例 3 - 単純な Head 要求 (戻りページなし)

```
REQUEST DOCUMENT FROM "http://pcnatweb"  
RESPONSE #rc
```

### 例 4 - 単純な Post 要求 (デフォルト)

```
REQUEST DOCUMENT FROM "http://pcnatweb/cgi-bin/nwwcgi.exe/sysweb/nat-env"  
WITH  
  DATA  
  NAME 'XMLData'          VALUE #Queryxml  
  NAME 'repServerName' VALUE 'NT2'  
RETURN  
  PAGE #Resultxml  
RESPONSE #rc
```

**例 5 - 単純な Put 要求 (すべてのデータを処理)**

```
REQUEST DOCUMENT FROM "http://pcnatweb/test.txt"  
WITH  
  DATA ALL      #document  
RETURN  
  PAGE #Resultxml  
RESPONSE #rc
```





# 109      RESET

---

▪ 機能 .....	752
▪ 構文説明 .....	752
▪ 例 .....	753

RESET [INITIAL] *operand1* ...

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ADD](#) | [COMPRESS](#) | [COMPUTE](#) | [DIVIDE](#) | [EXAMINE](#) | [MOVE](#) | [MOVE ALL](#) | [MULTIPLY](#) | [SEPARATE](#) | [SUBTRACT](#)

関連機能グループ：「[算術演算とデータ移動操作](#)」

## 機能

RESET ステートメントは、フィールドの値をリセットするために使用します。

- RESET (INITIAL なし) は、指定された各フィールドの内容をフォーマットに依存したデフォルトの初期値に設定します。
- **RESET INITIAL** は、指定された各フィールドを [DEFINE DATA](#) ステートメントのフィールド定義に従った初期値に設定します。DEFINE DATA ステートメントで **INIT** 節を指定せずに宣言したフィールドに対して、RESET INITIAL は RESET (INITIAL なし) と同じ効果を持ちます。



### 注意:

1. DEFINE DATA ステートメントで **CONSTANT** 節を指定して宣言したフィールドは、内容を変更できないので、RESET ステートメントで参照できません。
2. レポートモードでは、プログラムに [DEFINE DATA LOCAL](#) ステートメントが含まれていなければ、RESET ステートメントを使用して変数を定義することもできます。

## 構文説明

オペランド定義テーブル:

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A G M A U N P I F B D T L C G O		可	可

構文要素の説明：

<p><b>RESET</b> <i>operand1</i></p>	<p>空値にリセット：</p> <p>RESET (INITIALなし) は、指定された各フィールドの内容 (<i>operand1</i>) をデフォルトの初期値にリセットします。</p> <p><i>operand1</i>がダイナミック変数の場合、RESETステートメントの実行時に変数に含まれている長さの空値にリセットされます。ダイナミック変数の現在の長さは、システム変数 *LENGTH を使用して確認できます。</p> <p>ダイナミック変数に関する全般的な情報については、「ラージ変数／フィールドとダイナミック変数／フィールド」を参照してください。</p>
<p><b>RESET</b> <b>INITIAL</b> <i>operand1</i></p>	<p>初期値にリセット：</p> <p>RESET INITIAL は、指定された各フィールド (<i>operand1</i>) を DEFINE DATA ステートメントのフィールド定義に従った初期値に設定します。</p> <ul style="list-style-type: none"> <li>■ DEFINE DATA ステートメントに INIT 値を指定していない場合、フィールドはそのフォーマットに基づいてデフォルトの初期値で初期化されます。</li> <li>■ ダイナミック変数を使用する場合、初期値が定義されていなければ、*LENGTH はゼロに設定されます。</li> <li>■ RESET INITIAL を配列に適用する場合、DEFINE DATA ステートメントで定義した配列全体に適用する必要があります。配列の個々のオカレンスに RESET INITIAL を適用することはできません。</li> <li>■ 再定義で発生したフィールドに RESET INITIAL は使用できません。</li> <li>■ RESET INITIAL はダイナミック変数に適用されます。</li> <li>■ RESET INITIAL はデータベースフィールドには適用できません。</li> </ul>

例

```

** Example 'RSTEX1': RESET (with/without INITIAL)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
1 #BINARY (B4) INIT <1>
1 #INTEGER (I4) INIT <5>
1 #NUMERIC (N2) INIT <25>
END-DEFINE
*
LIMIT 1
READ EMPLOY-VIEW
/*
WRITE NOTITLE 'VALUES BEFORE RESET STATEMENT:'

```

```
WRITE / '=' NAME '=' #BINARY '=' #INTEGER '=' #NUMERIC
/*
RESET NAME #BINARY #INTEGER #NUMERIC
/*
WRITE /// 'VALUES AFTER RESET STATEMENT:'
WRITE / '=' NAME '=' #BINARY '=' #INTEGER '=' #NUMERIC
/*
RESET INITIAL #BINARY #INTEGER #NUMERIC
/*
WRITE /// 'VALUES AFTER RESET INITIAL STATEMENT:'
WRITE / '=' NAME '=' #BINARY '=' #INTEGER '=' #NUMERIC
/*
END-READ
END
```

プログラム **RSTEX1** の出力：

VALUES BEFORE RESET STATEMENT:

```
NAME: ADAM                #BINARY: 00000001 #INTEGER:          5 #NUMERIC:
25
```

VALUES AFTER RESET STATEMENT:

```
NAME:                    #BINARY: 00000000 #INTEGER:          0 #NUMERIC:
0
```

VALUES AFTER RESET INITIAL STATEMENT:

```
NAME:                    #BINARY: 00000001 #INTEGER:          5 #NUMERIC:
25
```

# 110 RESIZE

---

▪ 機能 .....	756
▪ 構文説明 .....	756

```
RESIZE { dynamic-clause } [GIVING operand5]  
      { array-clause }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[EXPAND](#) | [REDUCE](#)

関連機能グループ：「[ダイナミック変数または X-array のメモリ管理制御](#)」

## 機能

RESIZE ステートメントは、次を調整するために使用します。

- ダイナミック変数 (*dynamic-clause*) のサイズ、または
- X-array (*array-clause*) のオカレンス数

詳細については、『[プログラミングガイド](#)』の次のセクションを参照してください。

[ダイナミック変数の使用](#)

[ダイナミック変数のメモリスペースの割り当て／解放](#)

[X-array](#)

[X-Group 配列のストレージ管理](#)

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A	A U B	不可	不可
<i>operand2</i>	C S	I	不可	不可
<i>operand3</i>	A G	A N P I F B D T L C G O	可	不可
<i>operand4</i>	C S	N P I	不可	不可
<i>operand5</i>	S	I4	不可	可

構文要素の説明：

<i>dynamic-clause</i>	RESIZE DYNAMIC ステートメントは、ダイナミック変数 ( <i>operand1</i> ) の現在割り当てられているストレージについて、割り当てられている長さを、 <i>operand2</i> で指定した値に調整します。詳細については、後述の「 <a href="#">Dynamic 節</a> 」を参照してください。
<i>operand1</i>	<i>operand1</i> は、長さが調整されるダイナミック変数です。
<i>operand2</i>	<i>operand2</i> は、ダイナミック変数の新しい長さを指定するために使用します。値として、負ではない整数の定数または Integer4 (I4) タイプの変数を指定する必要があります。
<i>array-clause</i>	RESIZE ARRAY ステートメントは、X-array ( <i>operand3</i> ) のオカレンス数を、( <i>dim[,dim[,dim]]</i> ) で指定した上下限に調整します。詳細については、以下の「 <a href="#">Array 節</a> 」を参照してください。
<i>operand3</i>	<i>operand3</i> は X-array です。X-array のオカレンスを拡張または削減できます。配列のインデックス表記はオプションです。各次元でインデックス表記として使用できるのは、全範囲を示す表記 * のみです。
<i>dim</i> <i>operand4</i>	X-array 拡張の上下限表記 ( <i>operand4</i> またはアスタリスク) は、ここで指定します。現在の上下限の値を使用する必要がある場合は、 <i>operand4</i> の代わりにアスタリスク (*) を指定する必要があります。詳細については、後述の「 <a href="#">次元</a> 」を参照してください。
<b>GIVING</b> <i>operand5</i>	GIVING 節を指定しない場合は、エラー発生時に Natural ランタイムエラー処理がトリガされます。  GIVING 節を指定した場合は、 <i>operand5</i> に、エラー発生時は Natural メッセージ番号、成功時はゼロが含まれます。

## Dynamic 節

```
[SIZE OF] DYNAMIC [VARIABLE] operand1 TO operand2
```

RESIZE DYNAMIC ステートメントは、ダイナミック変数 (*operand1*) の割り当てられている長さを、*operand2* で指定した値に調整します。

RESIZE ステートメントを使用した場合、現在割り当てられているストレージサイズが、増減のどちらが必要であるかに関係なく、要求された値に調整されます。

## Array 節

```
[AND RESET][OCCURRENCES OF] ARRAY operand3 TO(dim[,dim[,dim]])
```

RESIZE ARRAY ステートメントは、X-array (*operand3*) のオカレンス数を、(*dim[,dim[,dim]]*) で指定した上下限に調整します。

RESET オプションは、サイズ変更した X-array のすべてのオカレンスをデフォルトのゼロ値にリセットします。デフォルト (RESET オプションなし) では、実際の値は保持され、サイズ変更した (新しい) オカレンスがリセットされます。

RESIZE ステートメントで使用される上限または下限は、配列に定義された対応する上限または下限と正確に同じである必要があります。

例：

```
DEFINE DATA LOCAL
1 #a(I4/1:*)
1 #g(1:*)
  2 #ga(I4/1:*)

1 #i(i4)
END-DEFINE
...

*/ resizing #a (1:10)
RESIZE ARRAY #a TO (1:10)          /* #a is resized to
RESIZE ARRAY #a TO (*:10)          /* 10 occurrences.

/* resizing #ga (1:10,1:20)
RESIZE ARRAY #g TO (1:10)          /* 1st dimension is set to (1:10)
RESIZE ARRAY #ga TO (*:*,1:20)    /* 1st dimension is dependent and
                                  /* therefore kept with (*:*)
                                  /* 2nd dimension is set to (1:20)

RESIZE ARRAY #a TO (5:10)          /* This is rejected because the lower index
                                  /* must be 1 or *
RESIZE ARRAY #a TO (#i:10)        /* This is rejected because the lower index
                                  /* must be 1 or *

RESIZE ARRAY #ga TO (1:10,1:20)    /* (1:10) for the 1st dimension is rejected
                                  /* because the dimension is dependent and
                                  /* must be specified with (*:*)
```

詳細については、次を参照してください。

- X-array のストレージ管理
- X-Group 配列のストレージ管理



## 次元

$$\left\{ \begin{array}{c} operand4 \\ * \end{array} \right\} : \left\{ \begin{array}{c} operand4 \\ * \end{array} \right\}$$

X-array 拡張の上下限表記 (*operand4* またはアスタリスク) は、ここで指定します。現在の上下限の値を使用する必要がある場合は、*operand4* の代わりにアスタリスク (\*) を指定できます。\*. \* の代わりに単一のアスタリスクを指定できます。

次元数 (*dim*) は、X-array (1、2、または 3) と正確に一致している必要があります。

指定した次元のオカレンス数が、現在割り当てられているオカレンス数より小さい場合は、対応する次元のオカレンス数は変更されません。



# 111      RETRY

---

▪ 機能 .....	762
▪ 制限事項 .....	762
▪ 例 .....	762

RETRY

関連ステートメント：**ACCEPT/REJECT** | **AT BREAK** | **AT START OF DATA** | **AT END OF DATA** | **BACKOUT TRANSACTION** | **BEFORE BREAK PROCESSING** | **DELETE** | **END TRANSACTION** | **FIND** | **GET** | **GET SAME** | **GET TRANSACTION DATA** | **HISTOGRAM** | **LIMIT** | **PASSW** | **PERFORM BREAK PROCESSING** | **READ** | **STORE** | **UPDATE**

関連機能グループ：「データベースへのアクセスと更新」

## 機能

RETRY ステートメントは、ON ERROR ステートメントブロック内でのみ使用できます（ON ERROR ステートメントを参照）。他ユーザー用にホールド状態になっているレコードに再度アクセスするときに使用します。

レコードが他ユーザー用にすでにホールド状態になっている場合、Naturalはエラーメッセージ 3145を発行します。セッションパラメータ WH（ホールド状態でのレコードの待機）も参照してください。

RETRY ステートメントは、エラー 3145の原因となっているオブジェクト内に指定します。

レコードのホールドロジックの詳細については、『プログラミングガイド』の「レコードホールドロジック」セクションを参照してください。

## 制限事項

このステートメントは、Adabas データベースにアクセスするためにのみ使用できます。

## 例

```
** Example 'RTYEX1': RETRY
**
** CAUTION: Executing this example will modify the database records!
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
*
1 #RETRY (A1) INIT <' '>
END-DEFINE
*
```

```
FIND EMPLOY-VIEW WITH NAME = 'ALDEN'  
/*  
DELETE  
END TRANSACTION  
/*  
ON ERROR  
  IF *ERROR-NR = 3145  
    INPUT NO ERASE 10/1  
      'RECORD IS IN HOLD' /  
      'DO YOU WISH TO RETRY?' /  
      #RETRY '(Y)ES OR (N)O?'  
    IF #RETRY = 'Y'  
      RETRY  
    ELSE  
      STOP  
    END-IF  
  END-IF  
END-ERROR  
/*  
AT END OF DATA  
  WRITE NOTITLE *NUMBER 'RECORDS DELETED'  
END-ENDDATA  
END-FIND  
*  
END
```



# 112 RUN

---

■ 機能 .....	766
■ 構文説明 .....	766
■ ダイナミックなソーステキストの作成／実行 .....	767
■ 例 .....	768

RUN [REPEAT] *operand1* [*operand2* [(*parameter*)]] ... 40

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[プログラムおよびルーチンの呼び出し](#)」

## 機能

RUN ステートメントは、Natural システムファイルから Natural ソースプログラムを読み込んで実行するとき 사용합니다。

Natural RPC については、『[Natural リモートプロシージャコール \(RPC\)](#)』ドキュメントの「[サーバーに対する Natural ステートメントの注意事項](#)」を参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A	可	不可
<i>operand2</i>	C S A G	A U N P I F B D T L G	可	不可

構文要素の説明：

<b>REPEAT</b>	<p>RUN REPEAT は、複数の出力画面が (<b>INPUT</b> ステートメントによって) 生成されるような場合でも、プログラムの実行が終了するまで、ユーザーに入力を促さないようにします。</p> <p>この機能は、各画面でユーザーが応答せずにプログラムで複数画面の情報を表示するような場合に使用できます。</p>
<i>operand1</i>	<p>プログラム名：</p> <p><i>operand1</i> として、プログラムの名前を英数字定数または英数字変数の内容として指定できます。変数を使用する場合は、長さは 8 文字長で指定する必要があります。</p> <p>プログラムは、現在のライブラリまたは連結されたライブラリ (デフォルトの STEPLIB は SYSTEM) に保存できます。プログラムが見つからない場合、エラーメッセージが発行されます。</p> <p>プログラムはソースプログラムワークエリア内に読み込まれ、エリア内にあるソースプログラムに置き換わります。</p>
<i>operand2</i>	<p>パラメータ：</p>



	<p>RUN ステートメントは、実行するプログラムに対してパラメータを渡すためにも使用できます。パラメータは任意のフォーマットで指定できます。パラメータは、対応する INPUT フィールドに適したフォーマットに変換されます。すべてのパラメータは、Natural スタックの最上位に配置されます。</p> <p>パラメータは <b>INPUT</b> ステートメントで読み込まれます。最初の INPUT ステートメントが発行されると、INPUT ステートメントで指定されたフィールドに全パラメータが挿入されます。INPUT ステートメントには数字フォーマットで定義されたパラメータフィールドに対する符号指定 (SG=ON) が必要です。</p> <p>次の INPUT ステートメントで読み取られる以上のパラメータが渡されると、余分のパラメータは無視されます。パラメータの数はシステム変数 *DATA で取得できます。</p> <p><b>注意:</b> <i>operand2</i> が時刻変数 (フォーマット T) の場合は、変数内容のうち時刻コンポーネントのみが渡され、日付コンポーネントは渡されません。</p>
<i>parameter</i>	<p><i>operand2</i> が日付変数の場合は、この変数に対する <i>parameter</i> として、セッションパラメータ DF (『パラメータリファレンス』を参照) を指定できます。</p>

## ダイナミックなソーステキストの作成／実行

ソースまたはその一部をダイナミックに作成したプログラムは、RUN ステートメントでダイナミックにコンパイルおよび実行できます。

ダイナミックなソーステキスト作成を実行するには、グローバル変数にソーステキストを挿入し、その後、ソーステキストでそのグローバル変数を参照します。このとき、変数名の先頭文字のプラス記号 (+) をアンパサンド (&) に置き換えて参照します。グローバル変数の内容は、プログラムが RUN ステートメントで呼び出されたときに、ソーステキストとして解釈されます。

インデックス付きグローバル変数を、RUN ステートメントを介して呼び出されるプログラム内で使用しないでください。

グローバル変数にコメントや **INCLUDE** ステートメントを入れることはできません。

## 例

**RUN** ステートメントを含むプログラム：

```

** Example 'RUNEX1': RUN (with dynamic source program creation)
*****
DEFINE DATA
GLOBAL
  USING RUNEXGDA
LOCAL
1 #NAME (A20)
1 #CITY (A20)
END-DEFINE
*
INPUT 'Please specify the search values:' //
      'Name:' #NAME /
      'City:' #CITY
*
RESET +CRITERIA      /* defined in GDA 'RUNEXGDA'
*
IF #NAME = ' ' AND #CITY = ' '
  REINPUT 'Enter at least 1 value'
END-IF
*
IF #NAME NE ' '
  COMPRESS 'NAME' ' '=' #NAME '' INTO +CRITERIA LEAVING NO
END-IF
IF #CITY NE ' '
  IF +CRITERIA NE ' '
    COMPRESS +CRITERIA 'AND' INTO +CRITERIA
  END-IF
  COMPRESS +CRITERIA ' CITY =' #CITY '' INTO +CRITERIA LEAVING NO
END-IF
*
RUN 'RUNEXFND'
*
END

```

**RUN** ステートメントで実行されるプログラム **RUNEXFND**：

```

** Example 'RUNEXFND': RUN (program executed with RUN in RUNEX1)
*****
DEFINE DATA
GLOBAL
  USING RUNEXGDA
LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
2 NAME
2 CITY

```

```

END-DEFINE
*
* &CRITERIA filled with "NAME = 'xxxxx' AND CITY = 'xxxx'"
*
FIND NUMBER EMPLOY-VIEW WITH &CRITERIA
    RETAIN AS 'EMP-SET'
DISPLAY *NUMBER
*
END

```

グローバルデータエリア **RUNEXGDA** :

```

Global  RUNEXGDA  Library SYSEXSYN          DBID  10 FNR  32
Command                                         > +
I T L  Name                               F Length  Miscellaneous
All  -- ----->
      1 +CRITERIA                          A         80

```



# 113 SEND METHOD

---

■ 機能 .....	772
■ 構文説明 .....	772
■ 例 .....	775

```

SEND [METHOD] operand1 TO [OBJECT] operand2
[ WITH { operand3 [ (AD= { M O A } ) ] } ]
[ nX ]
[RETURN operand4]
[GIVING operand5]
    
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CREATE OBJECT](#) | [DEFINE CLASS](#) | [INTERFACE](#) | [METHOD](#) | [PROPERTY](#)

関連機能グループ：「[コンポーネントベースプログラミング](#)」

## 機能

SEND METHOD ステートメントは、オブジェクトの特定メソッドを呼び出すために使用します。コンポーネントベースプログラミングの詳細については、『[プログラミングガイド](#)』の「[NaturalX](#)」セクションを参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
operand1	C S	A	可	不可
operand2	S		不可	不可
operand3	C S A G	A U N P I F B D T L C G O	可	不可
operand4	S A	A U N P I F B D T L C G O	可	不可
operand5	S	N I	可	不可

ローカルクラスのメソッドに渡すことができるのは、フォーマット C および G のみです。

## 構文要素の説明：

<i>operand1</i>	<p>メソッド名：</p> <p><i>operand1</i>は、<i>operand2</i>で指定されるオブジェクトによってサポートされるメソッドの名前です。</p> <p>メソッド名はクラスの異なるインターフェイスで同一にできるため、混同を避けるために、<i>operand1</i>のメソッド名をインターフェイス名で修飾することもできます。</p> <p>次の例では、オブジェクト #03 は、メソッド Start 付きのインターフェイス Iterate を持っています。次のステートメントが適用されます。</p> <pre>* Specifying only the method name. SEND 'Start' TO #03 * Qualifying the method name with the interface name. SEND 'Iterate.Start' TO #03</pre> <p>インターフェイス名が指定されていない場合、Natural はクラスのすべてのインターフェイスのメソッド名を検索します。メソッド名が複数のインターフェイスで見つかった場合、ランタイムエラーが発生します。</p>
<i>operand2</i>	<p>オブジェクトハンドル：</p> <p>メソッドコールが送信されるオブジェクトのハンドル。</p> <p><i>operand2</i>は、オブジェクトハンドル (HANDLE OF OBJECT) として定義する必要があります。オブジェクトがすでに存在する必要があります。</p> <p>メソッド内で現在のオブジェクトのメソッドを呼び出すには、システム変数 *THIS-OBJECT を使用してください。</p>
<i>operand3</i>	<p>メソッドに固有のパラメータ：</p> <p><i>operand3</i>として、メソッドに固有のパラメータを指定できます。</p> <p>次の例では、オブジェクト #03 は、パラメータ Pos 付きのメソッド PositionTo を持っています。メソッドは次の方法で呼び出されます。</p> <pre>SEND 'PositionTo' TO #03 WITH Pos</pre> <p>メソッドはオプションのパラメータを持つことができます。メソッドが呼び出される場合は、オプションのパラメータを指定する必要はありません。オプションパラメータを省略するには、プレースホルダ 1X を使用します。n オプションパラメータを省略するには、プレースホルダ nX を使用します。</p>


	<p>次の例では、オブジェクト #04 のメソッド SetAddress はパラメータ FirstName、MiddleInitial、LastName、Street、City を持ち、そのうち MiddleInitial、Street、City がオプションです。次のステートメントが適用されます。</p> <pre>* Specifying all parameters. SEND 'SetAddress' TO #04 WITH FirstName MiddleInitial LastName Street City * Omitting one optional parameter. SEND 'SetAddress' TO #04 WITH FirstName 1X LastName Street City * Omitting all optional parameters. SEND 'SetAddress' TO #04 WITH FirstName 1X LastName 2X</pre> <p>オプションでない（必須）パラメータを省略すると、ランタイムエラーが発生します。</p>						
<p><b>AD=</b></p>	<p>属性定義： <i>operand3</i> が変数の場合は、次のいずれかの方法でマークすることができます。</p> <table border="1" data-bbox="277 682 1385 982"> <tr> <td data-bbox="277 682 816 760">AD=O</td> <td data-bbox="821 682 1385 760">変更不可。セッションパラメータ AD=O を参照してください。</td> </tr> <tr> <td data-bbox="277 766 816 905">AD=M</td> <td data-bbox="821 766 1385 905">変更可。セッションパラメータ AD=M を参照してください。 これはデフォルト設定です。</td> </tr> <tr> <td data-bbox="277 911 816 982">AD=A</td> <td data-bbox="821 911 1385 982">入力のみ。セッションパラメータ AD=A を参照してください。</td> </tr> </table> <p><i>operand3</i> が定数の場合は、AD を明示的に指定することはできません。定数には常に AD=O が適用されます。</p>	AD=O	変更不可。セッションパラメータ AD=O を参照してください。	AD=M	変更可。セッションパラメータ AD=M を参照してください。 これはデフォルト設定です。	AD=A	入力のみ。セッションパラメータ AD=A を参照してください。
AD=O	変更不可。セッションパラメータ AD=O を参照してください。						
AD=M	変更可。セッションパラメータ AD=M を参照してください。 これはデフォルト設定です。						
AD=A	入力のみ。セッションパラメータ AD=A を参照してください。						
<p><b>nX</b></p>	<p>省略されるパラメータ：</p> <p>表記 <i>nX</i> を使用して、次の <i>n</i> パラメータを省略するように指定できます。例えば、1X は次のパラメータを省略し、3X は次の3つのパラメータを省略します。これは、次の <i>n</i> パラメータに、メソッドに渡す値がないことを意味します。</p> <p>Natural で実装されたメソッドについて、省略されるパラメータは、ダイアログの <b>DEFINE DATA PARAMETER</b> ステートメントでキーワード <b>OPTIONAL</b> を使用して定義する必要があります。OPTIONAL は、値を呼び出し側オブジェクトからこのようなパラメータに渡すこともできるということを意味します。</p>						
<p><b>RETURN</b> <i>operand4</i></p>	<p><b>RETURN 節：</b></p> <p>RETURN 節を省略し、メソッドが戻り値を持っている場合、戻り値は破棄されます。</p> <p>RETURN 節を指定した場合、<i>operand4</i> にはメソッドの戻り値が含まれます。メソッドの実行が失敗すると、<i>operand4</i> はその初期値にリセットされます。</p> <p><b>注意：</b> Natural で記述されたクラスについて、メソッドの戻り値は、メソッドのパラメータデータエリアに1つの追加パラメータを入力し、それを BY VALUE RESULT でマークすることによって定義されます（詳細については、「<b>PARAMETER 節</b>」セクションを参照）。したがって、Natural で記述され、戻り値を持つメソッドのパラメータデータエリアには常にメソッドパラメータの次に1つの追加フィールドが含まれます。この点は、Natural で記述されたクラスのメソッドを呼び出し、SEND ステートメントのメソッドのパラメータデータエリアを使用したいときに考慮します。</p>						



<b>GIVING</b>	<b>GIVING 節：</b>
<i>operand5</i>	<p>GIVING 節を指定しなかった場合、エラーの発生時に Natural ランタイムエラー処理が開始されます。</p> <p>GIVING 節を指定した場合、<i>operand5</i> に、エラーの発生時は Natural メッセージ番号、成功時はゼロが含まれます。</p>

## 例

次の図は、この例で使用されているプログラミングオブジェクトの概要を示しています。対応するソースコードとプログラム出力を下記に示します。



## プログラム METH01 - クラスと複数のメソッドを使用した CREATE OBJECT および SEND METHOD

```

** Example 'METH01':  CREATE OBJECT and SEND METHOD
**                    using a class and several methods (see METH*)
*****
DEFINE DATA
LOCAL
  USING METHA
LOCAL
1 L-STUDENT HANDLE OF OBJECT
1 #NAME      (A20)
1 #STREET    (A20)
1 #CITY      (A20)
1 #SUM       (I4)
1 #MULTI     (I4)
END-DEFINE
*
CREATE OBJECT L-STUDENT OF CLASS 'STUDENTS' /* see METHCL for class
*
L-STUDENT.<> := 'John Smith'
*
SEND METHOD 'INIT' TO L-STUDENT             /* see METHCL
      WITH #VAR1 #VAR2 #VAR3 #VAR4
*
SEND METHOD 'SUMMATION' TO L-STUDENT        /* see METHCL
      WITH #VAR1 #VAR2 #VAR3 #VAR4
*
SEND METHOD 'MULTIPLICATION' TO L-STUDENT  /* see METHCL
      WITH #VAR1 #VAR2 #VAR3 #VAR4
*
#NAME      := L-STUDENT.<>
#SUM       := L-STUDENT.<>
#MULTI     := L-STUDENT.<>
*
SEND METHOD 'ADDRESS' TO L-STUDENT         /* see METHCL
*
#STREET := L-STUDENT.<>
#CITY   := L-STUDENT.<>
*
*
WRITE 'Name   :' #NAME
WRITE 'Street:' #STREET
WRITE 'City   :' #CITY
WRITE ' '
WRITE 'The summation of      ' #VAR1 #VAR2 #VAR3 #VAR4
WRITE 'is' #SUM
WRITE 'The multiplication of' #VAR1 #VAR2 #VAR3 #VAR4
WRITE 'is' #MULTI
*
END

```

**METH01** で使用されるクラス定義 **METHCL** :

```
** Example 'METHCL': DEFINE CLASS (used by METH01)
*****
* Defining class STUDENTS for METH01
*
DEFINE CLASS STUDENTS
  OBJECT
    USING METH0          /* Object data for class STUDENTS
  /*
  INTERFACE STUDENT-ARITHMETICS
    PROPERTY FULL-NAME
      IS NAME
    END-PROPERTY
    PROPERTY SUM
    END-PROPERTY
    PROPERTY MULTI
    END-PROPERTY
*
  METHOD INIT
    IS METH02
    PARAMETER USING METHA
  END-METHOD
  METHOD SUMMATION
    IS METH03
    PARAMETER USING METHA
  END-METHOD
  METHOD MULTIPLICATION
    IS METH04
    PARAMETER USING METHA
  END-METHOD
  END-INTERFACE
*
  INTERFACE STUDENT-ADDRESS
    PROPERTY STUDENT-NAME
      IS NAME
    END-PROPERTY
    PROPERTY STREET
    END-PROPERTY
    PROPERTY CITY
    END-PROPERTY
*
  METHOD ADDRESS
    IS METH05
  END-METHOD
  END-INTERFACE
END-CLASS
END
```

クラス **METHCL** およびサブプログラム **METH02**、**METH03**、**METH04**、**METH05** で使用されるローカルデータエリア **METHO** (オブジェクトデータ) :

Local	METHO	Library	SYSEXSYN	DBID	10	FNR	32	
Command							> +	
I	T	L	Name	F	Length	Miscellaneous		
All	--	-----					----->	
	1		NAME	A	20			
	1		STREET	A	30			
	1		CITY	A	20			
	1		SUM	I	4			
	1		MULTI	I	4			

プログラム **METH01**、クラス **METHCL**、およびサブプログラム **METH02**、**METH03**、**METH04** で使用されるパラメータデータエリア **METHA** :

Parameter	METHA	Library	SYSEXSYN	DBID	10	FNR	32	
Command							> +	
I	T	L	Name	F	Length	Miscellaneous		
All	--	-----					----->	
	1		#VAR1	I	4			
	1		#VAR2	I	4			
	1		#VAR3	I	4			
	1		#VAR4	I	4			

サブプログラム **METH02** - プログラム **METH01** で使用されるメソッド **INIT** :

```

** Example 'METH02': Method INIT (used by METH01)
*****
DEFINE DATA
PARAMETER
  USING METHA
OBJECT
  USING METHO
END-DEFINE
*
* Method INIT of class STUDENTS
*
#VAR1 := 1
#VAR2 := 2
#VAR3 := 3
#VAR4 := 4
*
END

```

サブプログラム **METH03** - プログラム **METH01** で使用されるメソッド **SUMMATION** :

```

** Example 'METH03': Method SUMMATION (used by METH01)
*****
DEFINE DATA
PARAMETER
  USING METHA
OBJECT
  USING METHO
END-DEFINE
*
* Method SUMMATION of class STUDENTS
*
COMPUTE SUM = #VAR1 + #VAR2 + #VAR3 + #VAR4
END

```

サブプログラム **METH04** - プログラム **METH01** で使用されるメソッド **MULTIPLICATION** :

```

** Example 'METH04': Method MULTIPLICATION (used by METH01)
*****
DEFINE DATA
PARAMETER
  USING METHA
OBJECT
  USING METHO
END-DEFINE
*
* Method MULTIPLICATION of class STUDENTS
*
COMPUTE MULTI = #VAR1 * #VAR2 * #VAR3 * #VAR4
END

```

サブプログラム **METH05** - プログラム **METH01** で使用されるメソッド **ADDRESS** :

```

** Example 'METH05': Method ADDRESS (used by METH01)
*****
DEFINE DATA
  OBJECT USING METHO
END-DEFINE
*
* Method ADDRESS of class STUDENTS
*
IF NAME = 'John Smith'
  STREET := 'Oxford street'
  CITY   := 'London'
END-IF
END

```

プログラム **METH01** の出力：

Page 1 05-01-17 15:59:04


Name : John Smith  
Street: Oxford street  
City : London

The summation of	1	2	3	4
is	10			
The multiplication of	1	2	3	4
is	24			

# 114 SEPARATE

---

■ 機能 .....	782
■ 構文説明 .....	782
■ 例 .....	785



構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[COMPRESS](#) | [COMPUTE](#) | [EXAMINE](#) | [MOVE](#) | [MOVE ALL](#) | [RESET](#)

関連機能グループ：「[算術演算とデータ移動操作](#)」

## 機能

SEPARATE ステートメントは1つの英数字またはバイナリオペランドの内容を複数の英数字またはバイナリオペランド（または英数字やバイナリ配列の複数オカレンス）に分割するために使用します。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A U B	可	不可
<i>operand2</i>	C S	N P I B*	可	不可
<i>operand3</i>	C S	N P I B*	可	不可
<i>operand4</i>	S A G	A U B	可	可
<i>operand5</i>	S	A U B	可	可
<i>operand6</i>	C S	A U B	可	不可
<i>operand7</i>	S	N P I	可	可

\* *operand2* および *operand3* のフォーマット B は、4以下の長さでのみ使用できます。



## 構文要素の説明：

<i>operand1</i>	<p>ソースオペランド：</p> <p><i>operand1</i> は、内容を分割する英数字／バイナリの定数または変数です。</p> <p><i>operand1</i> 内の末尾の空白は、その値が処理される前に削除されます（デリミタ文字として空白を使用する場合でも同様です。<b>DELIMITERS</b> オプションも参照）。</p>
<b>SUBSTRING</b>	<p>通常、フィールドの内容全体が先頭から分割されます。</p> <p>SUBSTRING オプションでは、フィールドの一部だけを分割できます。SUBSTRING 節のフィールド名 (<i>operand1</i>) の後に、まず開始位置 (<i>operand2</i>)、次に分割するフィールド部分の長さ (<i>operand3</i>) を指定します。例えば、フィールド #A の値が CONTRAPTION の場合、SUBSTRING(#A,5,3) の内容は RAP を示します。</p> <p><b>注意:</b> <i>operand2</i> を省略すると、開始位置は 1 と見なされます。<i>operand3</i> を省略すると、長さはフィールドの開始位置から終わりまでと見なされます。</p>
<b>LEFT JUSTIFIED</b>	<p>このオプションでは、ターゲットオペランドからデリミタ文字と次の空白以外の文字の間に存在する先頭の空白を削除します。</p>
<i>operand4</i>	<p>ターゲットオペランド：</p> <p><i>operand4</i> は、ターゲットオペランドを表します。ターゲットオペランドとして配列が指定されている場合、分割された値は各オカレンスに送られます。</p> <p>ターゲットオペランドの個数は、<i>operand1</i> 内のデリミタ文字（末尾のデリミタ文字を含む）の個数に 1 を加えたものに相当します。</p> <p><i>operand4</i> が動的変数の場合、SEPARATE 操作によってその長さを変更できます。動的変数の現在の長さは、システム変数 *LENGTH を使用して確認できます。</p> <p>動的変数に関する全般的な情報については、「ラージ変数／フィールドと動的変数／フィールド」を参照してください。</p>
<b>IGNORE / REMAINDER</b> <i>operand5</i>	<p>分割されたソース値を受け取るターゲットフィールドが足りないと、該当するエラーメッセージが返されます。</p> <p>これを避けるために、次の 2 つのオプションを使用できます。</p> <ul style="list-style-type: none"> <li>■ <b>IGNORE</b> を指定すると、Natural はソース値を受け取るターゲットオペランドが足りない場合はそれを無視します。</li> <li>■ <b>REMAINDER</b> <i>operand5</i> を指定すると、ターゲットオペランドに挿入できなかったソース値は <i>operand5</i> に挿入されます。<i>operand5</i> の内容は、後続の SEPARATE ステートメントなどの処理に使用できます。</li> </ul> <p><b>例 3</b> も参照してください。</p>
<b>DELIMITERS</b>	<p>下記の「<b>デリミタオプション</b>」を参照してください。</p>

<b>WITH RETAINED DELIMITERS</b>	<p>通常、デリミタ文字自体はターゲットオペランドには移されません。</p> <p>RETAINED を指定すると、デリミタ（デフォルトのデリミタや空白、あるいは <i>operand6</i> で指定されたデリミタ）もターゲットオペランドに挿入されます。</p> <p>例：</p> <p>次の SEPARATE ステートメントでは、150 を #B、+ を #C、30 を #D に挿入しています。</p> <pre>... MOVE '150+30' TO #A SEPARATE #A INTO #B #C #D WITH RETAINED DELIMITER '+' ...</pre> <p><a href="#">例3</a> も参照してください。</p>
<b>GIVING NUMBER<sub>operand7</sub></b>	<p>GIVING NUMBER オプションでは、値（空白で返されたものも含む）が挿入されたターゲットオペランドの個数を <i>operand7</i> に返します。実際に得られる数はデリミタ文字の個数に 1 を加えたものです。</p> <p>IGNORE オプションを使用した場合、<i>operand7</i> に返される有効な最大数はターゲットオペランド (<i>operand4</i>) の個数です。</p> <p>REMAINDER オプションを使用した場合、<i>operand7</i> に返される有効な最大数はターゲットオペランド (<i>operand4</i>) の個数に <i>operand5</i> を加えたものです。</p>

デリミタオプション：

<b>[ WITH [RETAINED]</b>	<b>{ [ANY] DELIMITERS INPUT DELIMITERS DELIMITERS<sub>operand6</sub> }</b>
--------------------------	--

*operand1* 内のデリミタ文字は、値を分割する位置を示します。

- DELIMITERS オプションを省略した場合（または WITH ANY DELIMITERS を指定した場合）、空白、および英字や数字以外の任意の文字がデリミタ文字として扱われます。
- WITH INPUT DELIMITERS は、空白およびデフォルトの INPUT 区切り文字（セッションパラメータ ID で指定）がデリミタ文字として使用されることを示します。
- WITH DELIMITERS *operand6* は、指定した各文字 (*operand6*) がデリミタ文字として使用されることを示します。 *operand6* に含まれる末尾の空白は無視されます。

## 例

- 例 1-各種サンプル
- 例 2-配列の使用
- 例 3-REMAINDER/RETAINED オプションの使用

### 例 1-各種サンプル

```

** Example 'SEPEX1': SEPARATE
*****
DEFINE DATA LOCAL
1 #TEXT1   (A6) INIT <'AAABBB'>
1 #TEXT2   (A7) INIT <'AAA BBB'>
1 #TEXT3   (A7) INIT <'AAA-BBB'>
1 #TEXT4   (A7) INIT <'A.B/C,D'>
1 #FIELD1A (A6)
1 #FIELD1B (A6)
1 #FIELD2A (A3)
1 #FIELD2B (A3)
1 #FIELD3A (A3)
1 #FIELD3B (A3)
1 #FIELD4A (A3)
1 #FIELD4B (A3)
1 #FIELD4C (A3)
1 #FIELD4D (A3)
1 #NBT     (N1)
1 #DEL     (A5)
END-DEFINE
*
WRITE NOTITLE 'EXAMPLE A (SOURCE HAS NO BLANKS)'
SEPARATE #TEXT1 INTO #FIELD1A #FIELD1B GIVING NUMBER #NBT
WRITE      / '=' #TEXT1 5X '=' #FIELD1A 4X '=' #FIELD1B 4X '=' #NBT
*
WRITE NOTITLE /// 'EXAMPLE B (SOURCE HAS EMBEDDED BLANK)'
SEPARATE #TEXT2 INTO #FIELD2A #FIELD2B GIVING NUMBER #NBT
WRITE      / '=' #TEXT2 4X '=' #FIELD2A 7X '=' #FIELD2B 7X '=' #NBT
*
WRITE NOTITLE /// 'EXAMPLE C (USING DELIMITER ''-'')'
SEPARATE #TEXT3 INTO #FIELD3A #FIELD3B WITH DELIMITER '-'
WRITE      /      '=' #TEXT3 4X '=' #FIELD3A 7X '=' #FIELD3B
*
MOVE ',/' TO #DEL
WRITE NOTITLE /// 'EXAMPLE D USING DELIMITER' '=' #DEL
*
SEPARATE #TEXT4 INTO #FIELD4A #FIELD4B
                #FIELD4C #FIELD4D WITH DELIMITER #DEL
WRITE      /      '=' #TEXT4 4X '=' #FIELD4A 7X '=' #FIELD4B
                /
                19X '=' #FIELD4C 7X '=' #FIELD4D

```

## SEPARATE

```
*  
END
```

プログラム **SEPEX1** の出力 :

```
EXAMPLE A (SOURCE HAS NO BLANKS)
```

```
#TEXT1: AAABBB      #FIELD1A: AAABBB      #FIELD1B:           #NBT:  1
```

```
EXAMPLE B (SOURCE HAS EMBEDDED BLANK)
```

```
#TEXT2: AAA BBB      #FIELD2A: AAA          #FIELD2B: BBB        #NBT:  2
```

```
EXAMPLE C (USING DELIMITER '-')
```

```
#TEXT3: AAA-BBB      #FIELD3A: AAA          #FIELD3B: BBB
```

```
EXAMPLE D USING DELIMITER #DEL: ,/
```

```
#TEXT4: A.B/C,D      #FIELD4A: A.B          #FIELD4B: C  
                  #FIELD4C: D          #FIELD4D:
```

## 例 2 - 配列の使用

```
** Example 'SEPEX2': SEPARATE (using array variable)  
*****  
DEFINE DATA LOCAL  
1 #INPUT-LINE (A60) INIT <'VALUE1,  VALUE2,VALUE3'>  
1 #FIELD      (A20/1:5)  
1 #NUMBER     (N2)  
END-DEFINE  
*  
SEPARATE #INPUT-LINE LEFT JUSTIFIED INTO #FIELD (1:5)  
          GIVING NUMBER IN #NUMBER  
*  
WRITE NOTITLE #INPUT-LINE //  
              #FIELD (1) /  
              #FIELD (2) /  
              #FIELD (3) /  
              #FIELD (4) /  
              #FIELD (5) /  
              #NUMBER  
*  
END
```

プログラム **SEPEX2** の出力：

```
VALUE1, VALUE2,VALUE3
```

```
VALUE1
VALUE2
VALUE3
```

3

### 例 3 - REMAINDER/RETAINED オプションの使用

```
** Example 'SEPEX3': SEPARATE (with REMAINDER, RETAIN option)
*****
DEFINE DATA LOCAL
1 #INPUT-LINE (A60) INIT <'VAL1, VAL2, VAL3,VAL4'>
1 #FIELD      (A10/1:4)
1 #REM        (A30)
END-DEFINE
*
WRITE TITLE LEFT 'INP:' #INPUT-LINE /
              '#FIELD (1)' 13T '#FIELD (2)' 25T '#FIELD (3)'
              37T '#FIELD (4)' 49T 'REMAINDER'
/   '-----' 13T '-----' 25T '-----'
37T '-----' 49T '-----'
*
SEPARATE #INPUT-LINE INTO #FIELD (1:2)
          REMAINDER #REM WITH DELIMITERS ','
WRITE #FIELD(1) 13T #FIELD(2) 25T #FIELD(3) 37T #FIELD(4) 49T #REM
*
RESET #FIELD(*) #REM
SEPARATE #INPUT-LINE INTO #FIELD (1:2)
          IGNORE WITH DELIMITERS ','
WRITE #FIELD(1) 13T #FIELD(2) 25T #FIELD(3) 37T #FIELD(4) 49T #REM
*
RESET #FIELD(*) #REM
SEPARATE #INPUT-LINE INTO #FIELD (1:4) IGNORE
          WITH RETAINED DELIMITERS ','
WRITE #FIELD(1) 13T #FIELD(2) 25T #FIELD(3) 37T #FIELD(4) 49T #REM
*
RESET #FIELD(*) #REM
*
SEPARATE SUBSTRING(#INPUT-LINE,1,50) INTO #FIELD (1:4)
          IGNORE WITH DELIMITERS ','
WRITE #FIELD(1) 13T #FIELD(2) 25T #FIELD(3) 37T #FIELD(4) 49T #REM
*
END
```

プログラム **SEPEX3** の出力：

```
INP: VAL1, VAL2, VAL3,VAL4
#FIELD (1) #FIELD (2) #FIELD (3) #FIELD (4) REMAINDER
-----
VAL1 VAL2 VAL3,VAL4
VAL1 VAL2
VAL1 , VAL2 ,
VAL1 VAL2 VAL3 VAL4
```

# 115 SET CONTROL

---

■ 機能 .....	790
■ 構文説明 .....	790
■ 例 .....	790

SET CONTROLoperand1 ...

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

SET CONTROLステートメントは、端末コマンドをプログラム内から実行するために使用します。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
operand1	C S	A	可	不可

構文要素の説明：

operand1	<p>指定する端末コマンド：</p> <p>端末コマンドは、制御文字%なしでoperand1として指定し（デフォルト）、テキスト定数または英数字変数の内容として指定できます。</p> <p>端末コマンドの詳細については、『<a href="#">端末コマンド</a>』ドキュメントを参照してください。</p>
----------	--

## 例

- [例 1 - 小文字への変換](#)



---

- 例 2 - ハードコピー出力先の有効化

### 例 1 - 小文字への変換

```
...  
SET CONTROL 'L'  
...
```

小文字に変換します（端末コマンド %L に相当）。

### 例 2 - ハードコピー出力先の有効化

```
...  
SET CONTROL 'HDEST'...
```

ハードコピーの出力先を DEST にします（端末コマンド %Hdestination に相当）。



# 116 SET GLOBALS

---

■ 機能 .....	794
■ パラメータ .....	794
■ 例 .....	796

```
SET GLOBALS parameter=value ...
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

SET GLOBALS ステートメントは、セッションパラメータの値を設定するために使用します。

パラメータは、SET GLOBALS ステートメントを含むプログラムのコンパイル時または実行時に評価されます。どちらで評価されるかは、個々のパラメータによって異なります。

SET GLOBALS で指定したパラメータ設定は、後続の SET GLOBALS ステートメント（または GLOBALS システムコマンド）で上書きされない限り、Natural セッションが終了するまで有効です。SET GLOBALS ステートメントおよび GLOBALS システムコマンドには、同じ修正用パラメータが用意されています。両方とも同じ Natural セッション内で使用できます。GLOBALS コマンドで指定したパラメータ値は、新しい GLOBALS コマンドまたは SET GLOBALS ステートメントで上書きされるか、セッションが終了するか、またはユーザーが別のライブラリにログオンするまで有効です。

### 例外

下位プログラム（つまり、サブルーチン、サブプログラム、または FETCH RETURN で呼び出されるプログラム）内の SET GLOBALS ステートメントは、下位プログラムから呼び出し側オブジェクトに制御が返されるまで有効です。その後は、呼び出し側オブジェクト用に設定されたパラメータ値が再び適用されます。

## パラメータ

複数のパラメータを指定する場合、1つ以上の空白で区切る必要があります。パラメータは任意の順序で指定できます。[例](#)を参照してください。

SET GLOBALS ステートメントで設定可能なパラメータ	評価 (R = ランタイム、C = コンパイル時)
CC - 条件付きのプログラム実行	R
CF - 端末コマンドの文字	R
CPCVERR - コードページ変換エラー	R
DC - 小数点表記の文字	R
DC - 小数点表記の文字	R
DFOUT - 出力の日付フォーマット	R
DFSTACK - スタックの日付フォーマット	R
DFTITLE - デフォルトページタイトルの日付フォーマット	R

SET GLOBALS ステートメントで設定可能なパラメータ	評価 (R = ランタイム、C = コンパイル時)
DO - 出力データの表示順序	R
DU - ダンプ生成	R
EJ - ページ換え	R
FCDP - ダイナミックに保護されたフィールドの充填文字	R
FS - フォーマット指定	R
IA - INPUT 割り当て文字	R
ID - INPUT 区切り文字	R
IM - 入力モード	R
LE - エラー処理の制限	C
LS - 行サイズ	C
LT - レコードの読み取り制限	R
MT - 最大 CPU 時間	R
NC - Natural システムコマンドの使用	R
OPF - ヘルプルーチンによる保護されたフィールドの上書き	R
PD - NATPAGE ページデータセット	R
PM - 出力モード	C
PS - ページサイズ	RC
REINP - 不正データに対する内部的な REINPUT	R
SA - サウンド端末アラーム	R
SF - フィールド間の空白	C
TS - システムライブラリでプログラムからの出力を変換	R
WH - ホールド状態でのレコードの待機	R
ZD - ゼロ割り算のチェック	R
ZP - ゼロ出力	R

各セッションパラメータの詳細については、『パラメータリファレンス』を参照してください。

## 例

---

次の例では、SET GLOBALS ステートメントを使用して、1 行当たりの最大文字数を 74 文字に設定し、Natural プログラム内の処理ループで読み取ることができるデータベースレコードの数を 5000 に制限しています。

```
SET GLOBALS LS=74 LT=5000  
...
```

# 117 SET KEY

---

▪ 機能 .....	798
▪ 構文説明 .....	798
▪ キーのプログラム察知の有効化／キーの解除 .....	800
▪ コマンド／プログラムの割り当て .....	801
▪ 入力データの割り当て .....	802
▪ COMMAND OFF/ON .....	802
▪ HELP の割り当て .....	803
▪ DYNAMIC オプション .....	803
▪ DISABLED オプション .....	804
▪ 異なるプログラムレベルでの SET KEY ステートメント .....	804
▪ 名前の割り当て .....	805
▪ 例 .....	807

## 機能


---

SET KEY ステートメントは、次のタイプのキーに機能を割り当てるために使用します。

- ビデオ端末 PA (プログラムアテンション) キー
- PF (プログラムファンクション) キー
- CLEAR キー

SET KEY ステートメントを実行すると、Natural はプログラム実行中にキーの制御を受け取り、各キーに割り当てられた値を使用します。

Natural システム変数 \*PF-KEY は、最後に押されたキーを示します。

 **注意:** 機能が割り当てられていないキーを押すと、有効なキーを押すように促す警告メッセージが発行されるか、または Natural システム変数 \*PF-KEY に値 ENTR が設定されます。つまり、Natural は Enter キーが押されたときと同様に処理します。これは、Natural 管理者が Natural プロファイルパラメータ IKEY をどのように設定したかによります。PA キーおよび PF キーの処理は、Natural プロファイルパラメータ KEY (Natural 管理者が設定) の影響も受けます。

「ファンクションキーに基づいた処理」 (『プログラミングガイド』) も参照してください。

アプリケーションプログラミングインターフェイス: USR4005N 『ユーティリティ』ドキュメントの「SYSEXT-Natural アプリケーションプログラミングインターフェイス」を参照してください。

## 構文説明


---

このステートメントには、いくつかの構造を使用できます。


構文図で使用されている記号については、「[構文記号](#)」を参照してください。

構文 1 - すべてのキーに有効:






構文 2 - 個々のキーに有効：



構文 3 - 個々のキーに有効：




オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S	A	可	不可
<i>operand2</i>	C S	A U	可	不可
<i>operand3</i>	C S	A U	可	不可
<i>operand4</i>	C S	A U	可	不可

## キーのプログラム察知の有効化／キーの解除

キーをプログラム察知可能にするとは、現在アクティブなプログラムによるキーの問い合わせを可能にするということです。キーをプログラム察知可能にすると、キーの押下は Enter キーを押したときと同様の効果を持ちます。画面に入力されたすべてのデータがプログラムに転送されます。

 **注意:** PA キーや CLEAR キーをプログラム察知可能にしたときは、画面のデータは転送されません。

プログラム察知は、現在のプログラムの実行に対してのみ有効です。「異なるプログラムレベルでの **SET KEY ステートメント**」も参照してください。

例：

<b>SET KEY ALL</b>	このステートメントでは、すべてのキーをプログラム察知可能にします。各キーに割り当てられたすべての機能が上書きされます。
<b>SET KEY PF2</b> <b>SET KEY PF2=PGM</b>	これらの各ステートメントでは、PF2 をプログラム察知可能にします。
<b>SET KEY OFF</b>	このステートメントでは、すべてのキー設定を解除します。SET KEY OFF の実行後、システム変数 *PF-KEY に ENTR が設定されます。
<b>SET KEY ON</b>	このステートメントでは、設定が解除される前にすべてのキーに割り当てられていた機能を再び有効にし、プログラム察知可能だったすべてのキーを察知可能にします。
<b>SET KEY PF2=OFF</b>	このステートメントでは、PF2 の設定を解除します。SET KEY PF2=OFF の実行後、Natural システム変数 *PF-KEY に ENTR が設定されます（今まで PF2 が設定されていた場合）。
<b>SET KEY PF2=ON</b>	このステートメントでは、設定が解除される前に PF2 に割り当てられていた機能を再び有効にし、プログラム察知可能にします。PF2 に機能が割り当てられていなかった場合は、再びプログラム察知可能にします。

### キーのプログラム察知と \*PF-KEY の内容

次の例は、キーのプログラム察知とシステム変数 \*PF-KEY の内容の関係を示しています。

PF2がSET KEY PF2=PGMによりプログラム察知可能になっていて、後でINPUTステートメントが実行されることを前提としています。次の表は、ユーザーアクションと実行されるNaturalステートメントが\*PF-KEYの内容にどのように影響するかを示しています。

シーケンス	実行される Natural ステートメント / ユーザーアクション	*PF-KEY の内容
1	ユーザーによる PF2 の押下	PF2
2	SET KEY OFF	ENTR
3	SET KEY ON	PF2
4	SET KEY PF2=OFF	ENTR
5	SET KEY PF2=ON	PF2
6	SET KEY PF3=OFF	PF2

## コマンド / プログラムの割り当て

キーにコマンドまたはプログラム名を割り当てることができます。キーが押されると、現在のプログラムは終了し、そのキーに割り当てられているコマンド / プログラムがNaturalスタックを介して呼び出されます。コマンド / プログラムを割り当てる場合、コマンド / プログラムにパラメータを渡すこともできます（下記の3番目の例を参照）。


端末コマンドをキーに割り当てることもできます。キーが押されると、キーに割り当てられている端末コマンドが実行されます。

*operand2* を定数として指定する場合は、一重引用符で囲む必要があります。

例：

SET KEY PF4='SAVE'	コマンド SAVE が PF4 に割り当てられます。
SET KEY PF4=#XYX	変数 #XYZ の値が PF4 に割り当てられます。
SET KEY PF6='LIST MAP *'	コマンド LIST (LISTパラメータMAPおよび*を含む) が PF6 に割り当てられます。
SET KEY PF2='%%'	端末コマンド %% が PF2 に割り当てられます。

この割り当ては、別のSET KEYステートメントで上書きされるまで、別のアプリケーションにログオンするまで、またはNaturalセッションが終了するまで有効です。「異なるプログラムレベルでのSET KEYステートメント」も参照してください。

 **注意:** キーを通して呼び出されたプログラムを実行する前に、NaturalはBACKOUT TRANSACTIONステートメントを内部的に発行します。

## 入力データの割り当て

データ文字列 (*operand3*) をキーに割り当てることができます。キーが押されると、カーソルが現在置かれている入力フィールドにデータ文字列が挿入され、データは実行中のプログラムに転送されます (Enter キーが押された場合と同様)。

*operand3* を定数として指定する場合は、一重引用符で囲む必要があります。

例：

```
SET KEY PF12=DATA 'YES'
```

DATA 割り当ての整合性については、コマンド割り当てと同様のことが適用されます。つまり、別の SET KEY ステートメントで上書きされるまで、別のアプリケーションにログオンするまで、または Natural セッションが終了するまで有効です。「異なるプログラムレベルでの SET KEY ステートメント」も参照してください。

## COMMAND OFF/ON

COMMAND OFF では、キーに割り当てられている各機能 (コマンド、プログラム、データ) を一時的に解除できます。機能が割り当てられる前にキーがプログラム察知可能であった場合、COMMAND OFF によってそのキーが再び察知可能になります。

その後に COMMAND ON を使用して、割り当てられていた機能を再び有効にすることができます。

例：

SET KEY PF4=COMMAND OFF	PF4 に割り当てられている機能が一時的に解除されます。機能が割り当てられる前に PF4 がプログラム察知可能であった場合、再び察知可能になります。
SET KEY PF4=COMMAND ON	PF4 に割り当てられていた機能が再び有効になります。
SET KEY COMMAND OFF	すべてのキーに割り当てられている全機能が一時的に解除されます。機能が割り当てられる前にこれらのキーがプログラム察知可能であった場合、再び察知可能になります。
SET KEY COMMAND ON	すべてのキーに割り当てられていた全機能が再び有効になります。

SET KEY PF $nn$ ='' では、キーに割り当てられている機能を削除し、それと同時にキーのプログラム察知を解除できます。

## HELP の割り当て

HELP をキーに割り当てることができます。キーが押されると、カーソルが現在置かれているフィールドに割り当てられているヘルプルーチンが呼び出されます。

これはヘルプを呼び出すフィールドにヘルプ文字を入力した場合と同様です。ヘルプ文字は、Natural 管理者が設定した Natural プロファイルパラメータ HI によって決定されます。デフォルトは疑問符 (?) です。

例：

```
SET KEY PF1=HELP
```

HELP 割り当ての整合性については、プログラム察知と同様のことが適用されます。つまり、現在のプログラムの実行にのみ有効です。「異なるプログラムレベルでの **SET KEY** ステートメント」も参照してください。

## DYNAMIC オプション

SET KEY ステートメントで特定のキーを指定する代わりに、DYNAMIC オプションを変数 (*operand1*) とともに使用し、プログラム内でこの変数に値 (PF<sub>n</sub>、PA<sub>n</sub>、CLR) を割り当てることができます。これにより、どのキーに機能を割り当てるかをプログラムロジックによって指定できます。

例：

```
...  
IF ...  
    MOVE 'PF4' TO #KEY  
ELSE  
    MOVE 'PF7' TO #KEY  
END-IF  
...  
SET KEY DYNAMIC #KEY = 'SAVE'  
...
```

## DISABLED オプション

---

プッシュボタン、メニュー、ビットマップなどのグラフィカルユーザーインターフェイス (GUI) エレメントは PF キーとして実装されています。DISABLED オプションでは、PF キーに割り当てられた GUI エレメントを使用できないようにします。これにより、プッシュボタンとメニュー項目が灰色表示されます。

SET KEY PF $nn$ =DISABLED の設定を解除するには、SET KEY PF $nn$ =ON を使用します。

例：

SET KEY PF10=DISABLED	PF10 に割り当てられた GUI エレメントを使用できないようにします。
-----------------------	---------------------------------------

DISABLED オプションは処理ルール内でのみ使用できます。


## 異なるプログラムレベルでの SET KEY ステートメント

---

アプリケーション内の異なるレベルで SET KEY ステートメントが指定されている場合、次のことが適用されます。

- キーがプログラム察知可能な場合、各プログラムで別の SET KEY ステートメントが指定されるまで、下位レベルの（呼び出される）全プログラムにもこのプログラム察知が適用されます。制御が上位レベルのプログラムに返されると、上位レベルでの SET KEY 割り当てが再び有効になります。
- HELP キーとして定義されたキーについては、プログラム察知可能なキーと同様のことが適用されます。
- キーに機能（プログラム、コマンド、端末コマンド、データ文字列）が割り当てられると、この割り当ては（割り当てが行われたレベルに関係なく）すべての上位レベルおよび下位レベルで有効です。これは、そのキーに別の機能が割り当てられるか、キーがプログラム察知可能になるまで、別のアプリケーションにログオンするまで、または Natural セッションが終了するまで有効です。

## 異なるプログラムレベルでの SET KEY ステートメントの例



## 名前の割り当て

NAMED 節を使用して、キーに名前 ( *operand4* ) を割り当てることができます。名前は画面上の PF キー行に表示されます。これにより、キーに割り当てられている機能を識別できます。

```

? Help
. Exit
-----
Code ...: ? Library ...: *_____
          Object ....: *_____
          DBID .....: 0__ FILENR ...: 0__

Command ===>

```

Enter-PF1---	PF2---	PF3---	PF4---	PF5---	PF6---	PF7---	PF8---	PF9---	PF10--	PF11--	PF12---
Help	Exit	Last	Flip								Canc

PF キー行の表示はセッションパラメータ KD (『パラメータリファレンス』ドキュメントを参照) で有効になります。PF キー行の表示方法は端末コマンド %Y (『端末コマンド』ドキュメントを参照) で制御できます。

キーに割り当てる名前の最大長は 10 文字です。標準タブ PF キー行形式 (%YN) では、最初の 5 文字だけが表示されます。

*operand4* を定数として指定する場合は、一重引用符で囲む必要があります (下記の例を参照)。

機能を割り当てていないキーやプログラム察知可能でないキーに名前を割り当てることはできません。ただし、Enter キーだけは機能を割り当てなくても名前を割り当てることができます。

NAMED OFF では、プログラム察知可能なキーに割り当てた名前を削除できます。

例:

SET KEY ENTR NAMED 'EXEC'	Enter キーに EXEC という名前を割り当てます。
SET KEY PF3 NAMED 'EXIT'	PF3 をプログラム察知可能にし、PF3 に EXIT という名前を割り当てます。
SET KEY PF3 NAMED OFF	PF3 をプログラム察知可能にし、PF3 に割り当てられている名前を削除します。
SET KEY NAMED OFF	プログラム察知可能な各キーに割り当てられているすべての名前を削除します。
SET KEY PF4='AP1' NAMED 'APPL1'	PF4 にプログラム AP1 および名前 APPL1 を割り当てます。

標準タブ PF キー行形式 (%YN) を使用する場合、次のことが適用されます。

- キーにコマンド/プログラムを割り当てる際に NAMED 節を省略すると、コマンド/プログラム名が PF キー行に表示されます。コマンド/プログラム名が 5 文字を超過すると、CMND が表示されます。
- キーに入力データを割り当てる際に NAMED 節を省略すると、DATA が PF キー行に表示されます。
- (NAMED 節を使用して) 名前を Unicode フォーマットで PF キーに割り当てる場合、名前が対応するヘッダーに正しく配置されないことがあります。ただし、この問題は、*Natural Web I/O* インターフェイスを使用している場合、およびワイド文字に対してのみ発生します。この場合、順次 PF キー行形式 (%YS または %YP) を使用することをお勧めします。



順次PFキー行形式(%YSまたは%YP)を使用すると、名前を割り当てたキーのみがPFキー行に表示されます。つまり、キーにコマンド/プログラム/データを割り当てる際に NAMED 節を省略すると、キーはPFキー行に表示されません。

「ファンクションキー名に基づいた処理」(『プログラミングガイド』)も参照してください。

## 例

```

** Example 'SKYEX1': SET KEY
*****
DEFINE DATA LOCAL
1 #PF4 (A56)
END-DEFINE
*
MOVE 'LIST VIEW' TO #PF4
*
SET KEY PF1 PF2
SET KEY PF3 = 'MENU'
        PF4 = #PF4
        PF5 = 'LIST VIEW EMPLOYEES' NAMED 'Emp1'
*
FORMAT KD=ON
INPUT ////
        10X 'The following function keys are assigned:' //
        10X 'PF1: Function for PF1      ' /
        10X 'PF2: Function for PF2      ' /
        10X 'PF3: Return to MENU program' /
        10X 'PF4: LIST VIEW              ' /
        10X 'PF5: LIST VIEW EMPLOYEES   ' ///
*
IF *PF-KEY = 'PF1'
    WRITE 'Function for PF1 executed.'
END-IF
IF *PF-KEY = 'PF2'
    WRITE 'Function for PF2 executed.'
END-IF
*
END

```

プログラム **SKYEX1** の出力：

The following function keys are assigned:

PF1: Function for PF1

PF2: Function for PF2

PF3: Return to MENU program

PF4: LIST VIEW

PF5: LIST VIEW EMPLOYEES

# 118 SET TIME

---

■ 機能 .....	810
■ 例 .....	810

```
{ SET TIME }  
{ SETTIME }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

SET TIME (または SETTIME) ステートメントは、Natural システム変数 \*TIMD と組み合わせて、プログラムの特定のセクションを実行するために要した時間を計測するために使用します。

SET TIME ステートメントはプログラムの特定の位置に指定されます。\*TIMD の値は、SET TIME ステートメントの実行時点からの経過時間です。

\*TIMD には必ず SET TIME ステートメントへの参照を含める必要があります。そのためには、SET TIME ステートメントのソースコード行番号を使用するか、参照として使用できるラベルを SET TIME ステートメントに割り当てます。

## 例

```
** Example 'STIEX1': SETTIME  
*****  
DEFINE DATA LOCAL  
1 EMPLOY-VIEW VIEW OF EMPLOYEES  
  2 NAME  
END-DEFINE  
*  
ST. SETTIME  
WRITE 10X 'START TIME:' *TIME  
*  
READ (100) EMPLOY-VIEW BY NAME  
END-READ  
*  
WRITE NOTITLE 10X 'END TIME: ' *TIME  
WRITE          10X 'ELAPSED TIME TO READ 100 RECORDS'  
                '(HH:II:SS.T) : ' *TIMD (ST.) (EM=99:99:99'.'9)  
*  
END
```

---

プログラム **STIEX1** の出力：

```
START TIME: 16:39:07.6  
END TIME:   16:39:07.7  
ELAPSED TIME TO READ 100 RECORDS (HH:MM:SS.T) : 00:00:00.1
```



# 119 SET WINDOW

---

▪ 機能 .....	814
▪ 構文説明 .....	814
▪ 例 .....	814

```
SET WINDOW { 'window-name'
             OFF }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[DEFINE WINDOW](#) | [INPUT WINDOW='window-name'](#) | [REINPUT](#) | [SET WINDOW](#)

関連機能グループ：「[対話型処理用の画面生成](#)」

## 機能

SET WINDOW ステートメントは、ウィンドウをアクティブにしたり非アクティブにするために使用します。

SET WINDOW 'window-name' または INPUT WINDOW='window-name' ステートメントでは、現在アクティブなウィンドウを非アクティブにし、ステートメントで指定したウィンドウをアクティブにします。一度にアクティブにできるウィンドウは1つだけです。



**注意:** SIZE AUTO で定義されたウィンドウを SET WINDOW でアクティブにすると、ウィンドウがアクティブになる前の画面上のデータによってウィンドウのサイズが決定されます。

## 構文説明

SET WINDOW 'window-name'	指定したウィンドウをアクティブにします。つまり、そのウィンドウを非アクティブにするか、別のウィンドウをアクティブにするまで、後続のステートメントはすべてそのウィンドウを参照します。指定するウィンドウは <a href="#">DEFINE WINDOW</a> ステートメントで定義されている必要があります。
SET WINDOW OFF	現在アクティブなウィンドウを非アクティブにします。

## 例

[DEFINE WINDOW](#) ステートメントを参照してください。



# 120 SKIP

---

■ 機能 .....	816
■ 構文説明 .....	816
■ 例 .....	817

```
SKIP [(rep)] operand1 [LINES]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT END OF PAGE](#) | [AT TOP OF PAGE](#) | [CLOSE PRINTER](#) | [DEFINE PRINTER](#) | [DISPLAY](#) | [EJECT](#) | [FORMAT](#) | [NEWPAGE](#) | [PRINT](#) | [SUSPEND IDENTICAL SUPPRESS](#) | [WRITE](#) | [WRITE TITLE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

SKIP ステートメントは、1つまたは複数の空行を出力レポートに生成するために使用します。

『[ページタイトル、改ページ、空行](#)』（『[プログラミングガイド](#)』）も参照してください。

## 処理

SKIP ステートメントの実行によってページサイズを超過してしまった場合、超過分の行は無視されます（[AT TOP OF PAGE](#) ステートメントの場合を除く）。

SKIP ステートメントは、ページにすでに何らかの出力がある場合にのみ実行されます（[AT TOP OF PAGE](#) ステートメントからの出力を除く）。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	N P I	可	不可

構文要素の説明：

<i>(rep)</i>	<p>レポート指定：</p> <p>表記 <i>(rep)</i> を使用して、SKIP ステートメントを適用するレポートの ID を指定できます。</p> <p>範囲 0~31 の値、または <a href="#">DEFINE PRINTER</a> ステートメントを使用して割り当てた論理名を指定できます。</p> <p><i>(rep)</i> 指定がない場合、SKIP ステートメントは最初のレポート（レポート 0）に適用されます。</p>
--------------	---

	Natural で作成される出力レポートの形式を制御する方法については、『プログラミングガイド』の「データ出力制御」を参照してください。
<i>operand1</i>	<p>省略される行数：</p> <p><i>operand1</i> は、生成する空行の数（1～250）を表します。この数は、数値定数または数値変数の内容として指定できます。</p> <p><i>operand1</i> がレポートのページサイズを超過した場合、SKIP ステートメントは NEWPAGE 条件と同じ結果になります。</p>

## 例

```

** Example 'SKPEX1': SKIP
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 COUNTRY
  2 NAME
END-DEFINE
*
LIMIT 7
READ EMPL-VIEW BY CITY STARTING FROM 'W'
  AT BREAK OF CITY
    SKIP 2
  END-BREAK
  DISPLAY NOTITLE CITY (IS=ON) COUNTRY (IS=ON) NAME
  /*
END-READ
END

```

プログラム **SKPEX1** の出力：

CITY	COUNTRY	NAME
WASHINGTON	USA	REINSTEDT PERRY
WEITERSTADT	D	BUNGERT UNGER DECKER
WEST BRIDGFORD	UK	ENTWHISTLE

SKIP

---

WEST MIFFLIN

USA

WATSON

# 121 SORT

---

■ 機能 .....	820
■ 制限事項 .....	821
■ 構文説明 .....	821
■ SORT ステートメント処理 (3 段階) .....	823
■ 例 .....	824
■ 外部ソートプログラムの使用 .....	828

## ストラクチャードモード構文

```

END-ALL
[AND]
SORT [ THEM          ] [BY] { operand1 [ ASCENDING ] }... 10
      [ RECORDS      ]
      USING-clause
      [GIVE-clause]
      statement ...
END-SORT

```

\* ステートメントラベルを指定する場合、キーワード SORT の前、かつ END-ALL (および AND) の後に指定する必要があります。

## レポートモード構文

```

SORT [ THEM          ] [BY] { operand1 [ ASCENDING ] }... 10
      [ RECORDS      ]
      [USING-clause]
      [GIVE-clause]
      statement ...
[LOOP]

```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[FIND \(SORTED BY オプション付き\)](#)

関連機能グループ：「[ループ実行](#)」

## 機能

SORT ステートメントは、ソート処理を実行するために使用します。SORT ステートメントを実行すると、アクティブなすべての処理ループからのレコードがソートされます。

ソート処理には、Natural の内部ソートプログラムが使用されます。別の外部ソートプログラムを使用することもできます。

## 制限事項

- SORT ステートメントは、ソートするレコードを持つ処理ループと同じオブジェクト内に指定する必要があります。
- SORT ステートメントのネスト構造は許可されていません。
- ソートするレコードの合計長が 10240 バイトを超過しないようにする必要があります。
- ソート条件の数は 10 を超過しないようにする必要があります。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S	A N P I F B D T	不可	不可

構文要素の説明：

<b>END-ALL</b>	<p>ストラクチャードモードでは、SORT ステートメントの前に必ず END-ALL ステートメントを指定して、アクティブなすべての処理ループを終了する必要があります。SORT ステートメント自体では新しい処理ループを開始し、END-SORT ステートメントでループを終了する必要があります。</p> <p><b>注意:</b> レポートモードでは、SORT ステートメントはアクティブなすべての処理ループを終了させ、新しい処理ループを開始します。</p>
<i>operand1</i>	<p>ソート条件：</p> <p><i>operand1</i> は、ソート条件として使用するフィールド／変数を表します。1～10 個のデータベースフィールド（ディスクリプタおよび非ディスクリプタ）またはユーザー定義変数を指定できます。マルチプルバリューフィールドまたはピリオディックグループ内のフィールドを使用できます。グループまたは配列は指定できません。</p>
<b>ASCENDING</b>	デフォルトのソート順は昇順です。値を降順でソートする場合、DESCENDING を指定します。
<b>DESCENDING</b>	ASCENDING/DESCENDING は、各ソートフィールドに指定できます。
<b>USING</b>	下記の「 <b>USING 節</b> 」を参照してください。
<b>GIVE</b>	下記の「 <b>GIVE 節</b> 」を参照してください。
<b>END-SORT</b>	SORT ステートメントを終了するには、Natural 予約語 END-SORT を使用する必要があります。

## USING 節

USING 節では、中間ソートストレージに書き込むフィールドを指定します。レポーティングモードでは任意で、ストラクチャードモードでは必須です。ただし、これにより必要なメモリが軽減されるため、レポーティングモードでも使用することを強くお勧めします。

```
{ USING {operand2}... }
{ USING KEYS }
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand2</i>	S A	A N P I F B D T L C	不可	不可

構文要素の説明：

<b>USING</b> <i>operand2</i>	ソートキーフィールド ( <i>operand1</i> で指定) に加えて、中間ソートストレージに書き込む追加のフィールドを指定できます。
<b>USING KEYS</b>	<i>operand1</i> で指定したソートキーフィールドのみが中間ソートストレージに書き込まれます。


レポーティングモードの場合：USING 節を省略した場合、SORT ステートメントの前に開始された処理ループのすべてのデータベースフィールドおよび SORT ステートメントの前に定義されたすべてのユーザー定義変数が中間ソートストレージに書き込まれます。

ソートの実行終了後、中間ソートストレージに書き込まれなかったフィールドを参照した場合、フィールドの値はソート前のフィールドの最終値になります。

## GIVE 節

GIVE 節は、SORT ステートメントの第 1 段階で評価される Natural システム関数 (MAX、MIN など) を指定するために使用します。これらのシステム関数を第 3 段階で参照できます (「[SORT ステートメント処理](#)」を参照)。SORT ステートメントの後のシステム関数を参照するときには、\*AVER(SALARY) のように、前にアスタリスクを付ける必要があります。





オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
operand3	S A	*	可	不可

\* 使用するシステム関数に依存

構文要素の説明：

MAX   MIN   NMIN   COUNT   NCOUNT   OLD   AVER   NAVER   SUM   TOTAL	各システム関数の詳細については、『システム関数』ドキュメントを参照してください。
operand3	operand3はフィールド名です。
(NL=nn)	このオプションはAVER、NAVER、SUM、およびTOTALにのみ適用され、他のシステム関数では無視されます。  このオプションを使用して、システム関数の評価中の桁あふれを防ぐことができます。『システム関数』ドキュメントの「AVER、NAVER、SUM、TOTALでの桁あふれ」を参照してください。

## SORT ステートメント処理 (3 段階)

SORT ステートメントを含むプログラムは、3つの段階で実行されます。

### 第1段階 - ソートするレコードの選択

SORT ステートメントより前のステートメントが実行されます。USING 節に指定されたデータが中間ソートストレージに書き込まれます。

レポートモードでは、ソートの後で累計用に変数を使用する変数は、SORT ステートメントの前で定義しないでください。ストラクチャードモードでは、それらを **USING** 節に指定しないでください。中間ソートストレージに書き込まれるフィールドを累計用として使用することはできません。処理の第3段階で個々のレコードが読み直されるためです。各レコードについて、フィールドが読み直された個々のレコードの値で上書きされるため、累計機能として正しく処理されません。

中間ストレージに書き込まれるレコードの数は、処理ループの回数とループ当たりの処理レコードの数によって決まります。処理ループで SORT ステートメントが出現するたびに内部中間ストレージに 1 レコードが作成されます。ネスト構造のループの場合、内側のループが実行される場合にのみ、レコードは中間ストレージに書き込まれます。次の例で、内側の (FIND) ループで該当レコードが見つからない場合でも中間ストレージにレコードを書き込むためには、**FIND** ステートメントに **IF NO RECORDS FOUND** 節を指定しないでください。

```
READ ...
...
  FIND ...
...
END-ALL
SORT ...
  DISPLAY ...
END-SORT
...
```

## 第2段階 - レコードのソート

レコードがソートされます。

## 第3段階 - ソートされたレコードの処理

SORT ステートメントより後のステートメントが、指定のソート順で中間ストレージのすべてのレコードに対して実行されます。SORT ステートメントの後でデータベースフィールドを参照するには、適切なステートメントラベルまたは参照番号を使用して正しく参照する必要があります。

## 例

### ■ 例 1 - SORT

## ■ 例 2 - SORT

### 例 1 - SORT

```

** Example 'SRTEX1S': SORT (structured mode)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 CITY
  2 SALARY      (1:2)
  2 PERSONNEL-ID
  2 CURR-CODE  (1:2)
*
1 #AVG          (P11)
1 #TOTAL-TOTAL (P11)
1 #TOTAL-SALARY (P11)
1 #AVER-PERCENT (N3.2)
END-DEFINE
*
LIMIT 3
FIND EMPL-VIEW WITH CITY = 'BOSTON'
  COMPUTE #TOTAL-SALARY = SALARY (1) + SALARY (2)
  ACCEPT IF #TOTAL-SALARY GT 0
  /*
END-ALL
AND
SORT BY PERSONNEL-ID USING #TOTAL-SALARY SALARY(*) CURR-CODE(1)
  GIVE AVER(#TOTAL-SALARY)
  /*
AT START OF DATA
  WRITE NOTITLE '*' (40)
    'AVG CUMULATIVE SALARY:' *AVER (#TOTAL-SALARY) /
  MOVE *AVER (#TOTAL-SALARY) TO #AVG
END-START
COMPUTE ROUNDED #AVER-PERCENT = #TOTAL-SALARY / #AVG * 100
ADD #TOTAL-SALARY TO #TOTAL-TOTAL
/*
DISPLAY NOTITLE PERSONNEL-ID SALARY (1) SALARY (2)
  #TOTAL-SALARY CURR-CODE (1)
  'PERCENT/OF/AVER' #AVER-PERCENT
AT END OF DATA
  WRITE / '*' (40) 'TOTAL SALARIES PAID: ' #TOTAL-TOTAL
END-ENDDATA
END-SORT
*
END

```

プログラム **SRTEX1S** の出力：

PERSONNEL ID	ANNUAL SALARY	ANNUAL SALARY	#TOTAL-SALARY	CURRENCY CODE	PERCENT OF AVER
-----					
***** AVG CUMULATIVE SALARY:					41900
20007000	16000	15200	31200	USD	74.00
20019200	18000	17100	35100	USD	83.00
20020000	30500	28900	59400	USD	141.00
***** TOTAL SALARIES PAID:					125700

上記の例は次のように実行されます。

第 1 段階：

- EMPLOYEES ファイルから CITY=BOSTON のレコードが選択されます。
- SALARY の最初の 2 オカレンスがフィールド #TOTAL-SALARY に加算されます。
- #TOTAL-SALARY で、0 より大きいレコードのみが受け入れられます。
- レコードが中間ソートストレージに書き込まれます。データベース配列 SALARY（最初の 2 オカレンス）、CURR-CODE（第 1 オカレンス）、データベースフィールド PERSONNEL-ID、およびユーザー定義変数 #TOTAL-SALARY が中間ストレージに書き込まれます。
- #TOTAL-SALARY の平均値が評価されます。

第 2 段階：

- レコードがソートされます。

第 3 段階：

- ソートされた中間ストレージが読み取られます。
- AT START OF DATA 条件で #TOTAL-SALARY の平均値が表示されます。
- #TOTAL-SALARY が #TOTAL-TOTAL に加算され、フィールド PERSONNEL-ID、SALARY(1)、SALARY(2)、#AVER-PERCENT、および #TOTAL-SALARY が表示されます。
- AT END OF DATA 条件で変数 #TOTAL-TOTAL が出力されます。

レポートモードの例はライブラリ SYSEXRМ のプログラム **SRTEX1R** を参照してください。

**例 2 - SORT**

```
** Example 'SRTEX2': SORT
*****
DEFINE DATA LOCAL
1 VEHIC-VIEW VIEW OF VEHICLES
  2 MAKE
  2 YEAR
END-DEFINE
*
LIMIT 10
*
READ VEHIC-VIEW
END-ALL
SORT BY MAKE YEAR USING KEY
  DISPLAY NOTITLE (AL=15) MAKE (IS=ON) YEAR
  AT BREAK OF MAKE
  WRITE '-' (20)
  END-BREAK
END-SORT
END
```

プログラム **SRTEX2S** の出力：

MAKE	YEAR
FIAT	1980
	1982
	1984
-----	-----
PEUGEOT	1980
	1982
	1985
-----	-----
RENAULT	1980
	1980
	1982
	1982
-----	-----

## 外部ソートプログラムの使用

---

Natural では、ソート処理はデフォルトで Natural の内部ソートプログラムによって実行されます（上記参照）。ただし、外部ソートプログラムを使用することもできます。この外部ソートプログラムでは、Natural の内部ソートプログラムの代わりにソート処理を実行します。

使用する外部ソートプログラムは、Natural 管理者によって Natural パラメータモジュールのマクロ NTSORT で決定されます。外部ソートプログラムを使用するには、追加の JCL が必要です。詳細については、Natural 管理者に確認してください。

# 122      STACK

---

▪ 機能 .....	830
▪ 構文説明 .....	830
▪ 例 .....	832

```
STACK [TOP] { COMMANDoperand1 [operand2 [(parameter)]] ...
              [DATA] [FORMATTED] {operand2 [(parameter)]] ... }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[INPUT](#) | [RELEASE](#)

## 機能

STACK ステートメントは、Natural スタックに次のいずれかを挿入するために使用します。

- 実行する Natural プログラムまたは Natural システムコマンドの名前
- [INPUT](#) ステートメントの実行中に使用されるデータ

スタックの詳細については、「[プログラミングのその他のポイント](#)」の「スタック」（『[プログラミングガイド](#)』）を参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S A G N A		可	可
<i>operand2</i>	C S A G N A U N P I F B D T L G		可	可

構文要素の説明：

<b>TOP</b>	TOP を指定した場合、データ／プログラム／コマンドが Natural スタックの先頭に挿入されます。それ以外の場合、スタックの最後に挿入されます。
------------	--



	<p>例：次のステートメントでは、変数 #FIELDA の内容がデータとしてスタックの先頭に挿入されます。</p> <pre>STACK TOP #FIELDA</pre>
<b>DATA</b>	<p>DATA (デフォルト) では、スタックに挿入されたデータは <b>INPUT</b> ステートメントの入力データとして使用されます。</p> <p>データ内の区切り文字または INPUT 割り当て文字はデリミタとして処理されます。スタックのデータが INPUT ステートメントでどのように処理されるかについては、「<a href="#">Natural スタックデータの処理</a>」(INPUT ステートメントの説明内)を参照してください。</p> <p>例：次のステートメントでは、変数 #FIELD1 および #FIELD2 の内容がスタックに挿入されます。</p> <pre>MOVE 'ABC' TO #FIELD1 MOVE 'XYZ' TO #FIELD2 STACK #FIELD1 #FIELD2</pre> <p>これらの変数は、デリミタモードで Natural プログラム内の次の <b>INPUT</b> ステートメントにデータとして渡されます。</p> <pre>INPUT #FIELD1 #FIELD2</pre> <p><b>注意:</b> <i>operand2</i> が時刻変数 (フォーマット T) の場合は、変数内容のうち時刻コンポーネントのみがスタックに挿入され、日付コンポーネントは挿入されません。</p>
<b>FORMATTED</b>	<p>FORMATTED では、フィールド単位ですべてのデータを次の <b>INPUT</b> ステートメントに渡します。キー割り当てやデリミタ文字は解釈されません。</p> <p>例：</p> <p>次のステートメントでは、ABC,DEF が #FIELD1 に、XYZ が #FIELD2 に挿入されます。</p> <pre>MOVE 'ABC,DEF' TO #FIELD1 MOVE 'XYZ'      TO #FIELD2 STACK TOP DATA FORMATTED #FIELD1 #FIELD2 ... INPUT #FIELD1 #FIELD2</pre>

	<p>INPUT 区切り文字をコンマ（プロファイル/セッションパラメータ ID=,）にすると、次のステートメント（キーワード FORMATTED なし）により、ABC は #FIELD1 に、DEF は #FIELD2 に挿入されます。</p> <pre>MOVE 'ABC,DEF' TO #FIELD1 STACK TOP DATA #FIELD1 ... INPUT #FIELD1 #FIELD2</pre> <p><b>注意:</b> 渡されるデータにデリミタ文字、制御文字、または DBCS 文字が含まれている場合、これらの文字が不適切に解釈されないように FORMATTED オプションを使用してください。</p>
<b>COMMAND</b> <i>operand1</i>	<p>コマンド（またはプログラム名）をスタックに挿入するには、コマンド (<i>operand1</i>) をキーワード COMMAND に続けて指定します。Natural は、NEXT プロンプトを表示してユーザーに入力を促す代わりに、このコマンドを実行します。</p> <p>例：次のステートメントでは、コマンド RUN がスタックの先頭に挿入されます。通常では NEXT プロンプトが発行される時点で、Natural はこのコマンドを実行します。</p> <pre>STACK TOP COMMAND 'RUN'</pre>
<b>COMMAND</b> <i>operand1</i> <i>operand2</i> ...	<p>コマンド (<i>operand1</i>) とともにデータ (<i>operand2</i>) をスタックに挿入することもできます。これらのデータは、コマンド実行後の次の INPUT ステートメントで処理されます。</p> <p>コマンドとともにスタックに挿入されたデータは常にフォーマットされません。</p> <p><b>注意:</b> スタックに挿入されるデータに空の英数字フィールド（つまり、空白）が含まれている場合、これらの空白は値と値の間のデリミタとして解釈され、対応する INPUT ステートメントで正しく処理されません。そのため、コマンドとともに空の英数字フィールドをデータとしてスタックに挿入する場合、STACK ステートメントを2つ使用する必要があります。1つはデータをスタックに挿入するための STACK DATA <i>operand2</i> ...、もう1つはコマンドをスタックに挿入するための STACK COMMAND <i>operand1</i> です。</p>
<i>parameter</i>	<p><i>operand2</i> が日付変数の場合、この変数に対するパラメータとして、セッションパラメータ DF を指定できます。</p>

## 例

```
** Example 'STKEX1': STACK
*****
DEFINE DATA LOCAL
1 #CODE (A1)
END-DEFINE
*
INPUT //
  10X 'PLEASE SELECT COMMAND' //
  10X 'LIST VIEW          (V)' /
  10X 'LIST PROGRAM * (P)' /
```

```

10X 'TECH INFO      (T)' /
10X 'STOP           (.)' //
20X 'CODE:' #CODE
*
*
DECIDE ON FIRST #CODE
  VALUE 'V'
    STACK TOP DATA   'VIEW'
    STACK TOP COMMAND 'LIST'
  VALUE 'P'
    STACK TOP COMMAND 'LIST PROGRAM *'
  VALUE 'T'
    STACK TOP COMMAND 'LAST *'
    STACK TOP COMMAND 'TECH'
    STACK TOP COMMAND 'SYSPROD'
  VALUE '.'
    STOP
  NONE
    REINPUT 'PLEASE ENTER VALID CODE'
END-DECIDE
*
*
END

```

プログラム **STKEX1** の出力：

```

PLEASE SELECT COMMAND

LIST VIEW      (V)
LIST PROGRAM * (P)
TECH INFO      (T)
STOP           (.)

      CODE:P

```

コードを入力して確認した後：

```

16:46:28          ***** NATURAL LIST COMMAND *****          2005-01-19
User HTR          - LIST Objects in a Library -          Library SYSEXSYN

```

Cmd	Name	Type	S/C	SM	Version	User ID	Date	Time
---	*	P	*	*	*	*	*	*
___	ACREX1	Program	S/C	S	4.1.03	RKE	2004-11-11	16:32:37
___	ACREX2	Program	S/C	S	4.1.03	RKE	2005-01-05	10:29:51
___	ADDEX1	Program	S/C	S	4.1.03	RKE	2004-11-11	16:36:49
___	AEDEX1R	Program	S/C	R	4.1.03	RKE	2004-11-11	16:40:34
___	AEDEX1S	Program	S/C	S	4.1.03	RKE	2004-11-11	16:39:57
___	AEPEX1R	Program	S/C	R	4.1.03	RKE	2004-11-11	16:41:57
___	AEPEX1S	Program	S/C	S	4.1.03	RKE	2004-11-11	16:42:31
___	AEPEX2	Program	S/C	S	4.1.03	RKE	2004-11-11	16:43:37
___	ASDEX1R	Program	S/C	R	4.1.03	RKE	2004-11-11	17:00:21

## STACK

---

```
___ ASDEX1S   Program      S/C  S  4.1.03  RKE      2004-11-11  17:00:50
___ ASGEX1R   Program      S/C  R  4.1.03  RKE      2004-11-11  17:02:01
___ ASGEX1S   Program      S/C  S  4.1.03  RKE      2004-11-11  17:02:08
___ ATBEX1R   Program      S/C  R  4.1.03  RKE      2004-11-11  17:03:18
___ ATBEX1S   Program      S/C  S  4.1.03  RKE      2004-11-11  17:03:05
                                         14 Objects found
```

Top of List.

Command ==>

```
Enter-PF1---PF2---PF3---PF4---PF5---PF6---PF7---PF8---PF9---PF10--PF11--PF12---
      Help  Print Exit  Sort      --      -      +      ++      >      Canc
```

# 123 STOP

---

▪ 機能 .....	836
▪ 例 .....	836

STOP

## 機能

STOP ステートメントは、プログラムの実行を終了させ、コマンド入力プロンプトに戻すために使用します。

1 つ以上の STOP ステートメントを Natural プログラム内の任意の位置に挿入できます。

STOP ステートメントでは、プログラムの実行を即座に終了させます。サブルーチン内の STOP ステートメントの位置に関係なく、メインプログラムで指定されたページ終了条件は、STOP ステートメントの実行中に最後のページ終了処理のために呼び出されます。

Natural RPC については、『*Natural* リモートプロシージャコール (RPC) 』ドキュメントの「サーバーに対する *Natural* ステートメントの注意事項」を参照してください。

## 例

```

** Example 'STPEX1': STOP
*****
DEFINE DATA LOCAL
1 #CODE (A1)
END-DEFINE
*
INPUT //
  10X 'PLEASE SELECT COMMAND' //
  10X 'LIST VIEW      (V)' /
  10X 'LIST PROGRAM * (P)' /
  10X 'TECH INFO     (T)' /
  10X 'STOP          (.)' //
  20X 'CODE:' #CODE
*
*
DECIDE ON FIRST #CODE
  VALUE 'V'
    STACK TOP DATA      'VIEW'
    STACK TOP COMMAND    'LIST'
  VALUE 'P'
    STACK TOP COMMAND    'LIST PROGRAM *'
  VALUE 'T'
    STACK TOP COMMAND    'LAST *'
    STACK TOP COMMAND    'TECH'
    STACK TOP COMMAND    'SYSPROD'
  VALUE '.'
    STOP

```

```
NONE
  REINPUT 'PLEASE ENTER VALID CODE'
END-DECIDE
*
*
END
```

プログラム **STPEX1** の出力：

```
PLEASE SELECT COMMAND

LIST VIEW      (V)
LIST PROGRAM * (P)
TECH INFO      (T)
STOP           (.)

      CODE:
```





# 124 STORE

---

■ 機能 .....	840
■ データベース固有の考慮事項 .....	841
■ 構文説明 .....	841
■ 例 .....	843

## ストラクチャードモード構文

```
STORE [RECORD] [IN] [FILE] view-name
      [PASSWORD=operand1]
      [CIPHER=operand2]
      [
        [
          USING
          GIVING
        ] NUMBER operand3 ]
```

## レポートイングモード構文

```
STORE [RECORD] [IN] [FILE] view-name
      [PASSWORD=operand1]
      [CIPHER=operand2]
      [
        [
          USING
          GIVING
        ] NUMBER operand3 ]
      {
        [USING] SAME [RECORD] [AS] [STATEMENT [(r)]]
        [
          SET
          WITH
        ] [operand4=operand5]
        ...
      }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION DATA](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [UPDATE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

STORE ステートメントは、データベースにレコードを追加するために使用します。

## データベース固有の考慮事項

<b>Adabas</b>	Natural システム変数 *ISN には、STORE ステートメントの実行の結果として新しいレコードに割り当てられた Adabas ISN が含まれます。*ISN への後続の参照には、関連する STORE ステートメントのステートメント番号が含まれている必要があります。
<b>DL/I</b>	このステートメントは、セグメントオカレンスを追加するために使用できます。  データセットをプライマリーキーで定義する場合は、プライマリーキーフィールドに値を入力する必要があります。GSAM データベースの場合、レコードはデータベースの最後に追加する必要があります (GSAM 制限のため)。  Natural システム変数 *ISN は使用できません。
<b>SQL</b>	このステートメントは、SQL テーブルに行を追加するために使用できます。PASSWORD 節、CIPHER 節、および GIVING NUMBER 節は使用できません。STORE ステートメントは、SQL の INSERT ステートメントに相当します。  Natural システム変数 *ISN は使用できません。
<b>VSAM</b>	データセットをプライマリーキーで定義する場合は、プライマリーキーフィールドに値を入力する必要があります。  Natural システム変数 *ISN には、STORE ステートメントの実行の結果として新しいレコードに割り当てられた VSAM RBA/RRN が含まれます。*ISN への後続の参照には、関連する STORE ステートメントのステートメント番号が含まれている必要があります。  VSAM データベースの場合、Natural システム変数 *ISN は ESDS および RRDS ファイルに対してのみ使用できます。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A	可	不可
<i>operand2</i>	C S	N	可	不可
<i>operand3</i>	S	N P B*	不可	可
<i>operand4</i>	S A	A N P I F B D T L	不可	不可
<i>operand5</i>	C S A	A N P I F B D T L	可	不可

\* *operand3* のフォーマット B は、4 以下の長さでのみ使用できます。

## 構文要素の説明：

<i>view-name</i>	<p><i>view-name</i> として、ビューの名前を指定します。この名前は、<b>DEFINE DATA</b> ステートメント内、あるいはプログラム外のグローバルデータエリアまたはローカルデータエリアに事前に定義しておく必要があります。</p> <p>レポートモードでは、<b>DEFINE DATA</b> ステートメントを使用しない場合、<i>view-name</i> には <b>DDM</b> の名前も使用できます。</p>
<b>PASSWORD=</b> <i>operand1</i>	<p><b>PASSWORD</b> 節は、Adabas または VSAM データベースに対してのみ適用可能です。</p> <p>この節は、パスワード保護されたファイルのデータを更新するときのパスワード (<i>operand1</i>) の指定に使用します。パスワード (<i>operand1</i>) は、英数字定数または英数字変数として指定できます。最大 8 文字から成り、特殊文字または埋め込み空白は使用できません。パスワードを定数として指定する場合は、一重引用符で囲む必要があります。</p> <p>詳細については、<b>FIND</b> ステートメントおよび <b>PASSW</b> ステートメントを参照してください。</p>
<b>CIPHER=</b> <i>operand2</i>	<p><b>CIPHER</b> 節は、Adabas または VSAM データベースに対してのみ適用可能です。</p> <p>この節は、暗号化されたファイルのデータを更新するときのサイファキー (<i>operand2</i>) の指定に使用します。サイファキー (<i>operand2</i>) は、数値定数 (8 桁) またはユーザー定義変数 (フォーマット/長さ N8) として指定できません。</p> <p>詳細については、<b>FIND</b> ステートメントを参照してください。</p>
<b>USING NUMBER</b> <i>operand3</i>  または <b>GIVING NUMBER</b> <i>operand3</i>	<p>この節は、Adabas または VSAM データベースでのみ使用できます。VSAM データベースの場合、この節は VSAMRRDS に対してのみ有効です。この場合、ユーザー指定の RRN (相対レコード番号) は上記の ISN に対応します。</p> <p>この節は、ユーザー指定の Adabas ISN を持つレコードを保存するために使用します。指定した ISN を持つレコードがすでに存在しているときには、エラーメッセージが返され、ON ERROR 処理の指定がない場合、プログラムの実行は終了します。</p>
<b>SET/WITH</b> <i>operand4=operand5</i>	<p><b>SET/WITH</b> は、レポートモードで値を与えるフィールドを指定するために使用します。ファイルに定義されているフィールドで、<b>SET</b> 節で指定されないフィールドは、新しいレコード内で空値となります。</p> <p>この節は、<b>DEFINE DATA</b> ステートメントを使用している場合は許可されません。<b>STORE</b> ステートメントは <b>DEFINE DATA</b> ステートメント内の定義に従って常にビュー全体を参照するからです。</p> <p><b>DL/I-Specific</b> の考慮事項：</p> <p>可変長のセグメントは、<b>STORE</b> ステートメントで指定されたすべてのフィールドを含めるのに必要な最小長で保存されます。セグメント長は、<b>DBD</b> の <b>SEGM</b> マクロで指定された最小サイズより小さくなることはありません。値は、セグメントシーケンスフィールドのほか、上位のすべてのシーケンスフィールドに対して指定する必要があります。I/O (センシティブ) フィールドのみを指定できます。</p>

	<p>マルチプルバリューフィールドまたはピリオディックグループが可変長として定義されている場合、セグメントの最後で、STORE ステートメントに指定されたオカレンスのみがセグメントに書き込まれ、それに応じてセグメント長が定義されます。</p>
<b>USING SAME (r)</b>	<p>USING SAME は、レポーティングモードで使用され、STORE ステートメントが参照するステートメント (<b>FIND</b>、<b>GET</b>、<b>READ</b>) で読み込まれたフィールドと同じフィールドが新しいレコードとして追加されることを示します。ステートメント参照表記 (r) は、ソースコード行番号またはステートメントラベルとして指定できます。</p> <p>この節は、<b>DEFINE DATA</b> ステートメントを使用している場合は許可されません。STORE ステートメントは <b>DEFINE DATA</b> ステートメント内の定義に従って常にビュー全体を参照するからです。</p>

## 例

```

** Example 'STOEX1S': STORE (structured mode)
**
** CAUTION: Executing this example will modify the database records!
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
  2 MAR-STAT
  2 BIRTH
  2 CITY
  2 COUNTRY
*
1 #PERSONNEL-ID (A8)
1 #NAME (A20)
1 #FIRST-NAME (A15)
1 #BIRTH-D (D)
1 #MAR-STAT (A1)
1 #BIRTH (A8)
1 #CITY (A20)
1 #COUNTRY (A3)
1 #CONF (A1)
END-DEFINE
*
REPEAT
  INPUT 'ENTER A PERSONNEL ID AND NAME (OR ''END'' TO END)' //
        'PERSONNEL-ID : ' #PERSONNEL-ID //
        'NAME : ' #NAME /
        'FIRST-NAME : ' #FIRST-NAME
/*
/* VALIDATE ENTERED DATA

```

```
/*
IF #PERSONNEL-ID = 'END' OR #NAME = 'END'
  STOP
END-IF
IF #NAME = ' '
  REINPUT WITH TEXT 'ENTER A LAST-NAME' MARK 2 AND SOUND ALARM
END-IF
IF #FIRST-NAME = ' '
  REINPUT WITH TEXT 'ENTER A FIRST-NAME' MARK 3 AND SOUND ALARM
END-IF
/*
/* ENSURE PERSON IS NOT ALREADY ON FILE
/*
FIND NUMBER EMPL-VIEW WITH PERSONNEL-ID = #PERSONNEL-ID
IF *NUMBER > 0
  REINPUT 'PERSON WITH SAME PERSONNEL-ID ALREADY EXISTS'
  MARK 1 AND SOUND ALARM
END-IF
MOVE 'N' TO #CONF
/*
/* GET FURTHER INFORMATION
/*
INPUT
'ADDITIONAL PERSONNEL DATA'          ////
'PERSONNEL-ID           :' #PERSONNEL-ID (AD=IO) /
'NAME                   :' #NAME          (AD=IO) /
'FIRST-NAME             :' #FIRST-NAME   (AD=IO) ///
'MARITAL STATUS         :' #MAR-STAT     /
'DATE OF BIRTH (YYYYMMDD) :' #BIRTH      /
'CITY                   :' #CITY         /
'COUNTRY (3 CHARACTERS) :' #COUNTRY     //
'ADD THIS RECORD (Y/N)  :' #CONF        (AD=M)
/*
/* ENSURE REQUIRED FIELDS CONTAIN VALID DATA
/*
IF NOT (#MAR-STAT = 'S' OR = 'M' OR = 'D' OR = 'W')
  REINPUT TEXT 'ENTER VALID MARITAL STATUS S=SINGLE ' -
  'M=MARRIED D=DIVORCED W=WIDOWED' MARK 1
END-IF
IF NOT (#BIRTH = MASK(YYYYMMDD) AND #BIRTH = MASK(1582-2699))
  REINPUT TEXT 'ENTER CORRECT DATE' MARK 2
END-IF
IF #CITY = ' '
  REINPUT TEXT 'ENTER A CITY NAME' MARK 3
END-IF
IF #COUNTRY = ' '
  REINPUT TEXT 'ENTER A COUNTRY CODE' MARK 4
END-IF
IF NOT (#CONF = 'N' OR = 'Y')
  REINPUT TEXT 'ENTER Y (YES) OR N (NO)' MARK 5
END-IF
IF #CONF = 'N'
```

```
    ESCAPE TOP
  END-IF
/*
/*  ADD THE RECORD
/*
MOVE EDITED #BIRTH TO #BIRTH-D (EM=YYYYMMDD)
/*
EMPL-VIEW.PERSONNEL-ID := #PERSONNEL-ID
EMPL-VIEW.NAME         := #NAME
EMPL-VIEW.FIRST-NAME  := #FIRST-NAME
EMPL-VIEW.MAR-STAT    := #MAR-STAT
EMPL-VIEW.BIRTH       := #BIRTH-D
EMPL-VIEW.CITY        := #CITY
EMPL-VIEW.COUNTRY     := #COUNTRY
/*
STORE RECORD IN EMPL-VIEW
/*
END OF TRANSACTION
/*
WRITE NOTITLE 'RECORD HAS BEEN ADDED'
/*
END-REPEAT
END
```

プログラム **STOEX1S** の出力：

```
ENTER A PERSONNEL ID AND NAME (OR 'END' TO END)

PERSONNEL-ID : 90001100

NAME          : JONES
FIRST-NAME    : EDWARD
```

人事キーデータを入力および確定した後、追加の人事データフィールドが入力用に表示されます。

```
ADDITIONAL PERSONNEL DATA

PERSONNEL-ID          : 90001100
NAME                  : JONES
FIRST-NAME            : EDWARD

MARITAL STATUS        :
DATE OF BIRTH (YYYYMMDD) :
CITY                  :
COUNTRY (3 CHARACTERS) :

ADD THIS RECORD (Y/N) : N
```

レポートモードの例はライブラリ SYSEXRM のプログラム **STOEX1R** を参照してください。



# 125 SUBTRACT

---

▪ 機能 .....	848
▪ 構文説明 .....	848
▪ 例 .....	850

関連ステートメント：[ADD](#) | [COMPRESS](#) | [COMPUTE](#) | [DIVIDE](#) | [EXAMINE](#) | [MOVE](#) | [MOVE ALL](#) | [MULTIPLY](#) | [RESET](#) | [SEPARATE](#)

関連機能グループ：「[算術演算とデータ移動操作](#)」

## 機能

SUBTRACT ステートメントは、複数のオペランドの値を減算するために使用します。

データベースフィールドを結果フィールドとして使用した場合、SUBTRACT 演算ではプログラム内で使用される内部値のみが更新されます。データベース内のフィールドの値は変更されないまま維持されます。

## 構文説明

### 構文 1 - SUBTRACT

```
SUBTRACT [ROUNDED] operand1 ... FROM operand2
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S A N	N P I F D T	可	不可
<i>operand2</i>	S A M	N P I F D T	可	不可

構文要素の説明：

<i>operand1</i> FROM	オペランド：
<i>operand2</i>	<p><i>operand1</i>は被減数、<i>operand2</i>は減数です。したがって、ステートメントは次と等しくなります。</p> <p><math>\langle operand2 \rangle := \langle operand2 \rangle - \langle operand1 \rangle</math></p> <p>オペランドのフォーマットについては、「<a href="#">算術割り当てのルール</a>」の「<a href="#">フォーマット混合式のパフォーマンスについて</a>」（『<a href="#">プログラミングガイド</a>』）も参照してください。</p>
<a href="#">ROUNDED</a>	切り上げ：

キーワード `ROUNDED` を指定すると、結果は切り上げられます。切り上げについては、『プログラミングガイド』の「演算割り当てのルール」、「フィールドの切り捨てと切り上げ」を参照してください。

## 構文 2

```
SUBTRACT [ROUNDED] operand1 ... FROM operand2 GIVING operand3
```

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S A N	N P I F D T	可	不可
<i>operand2</i>	C S A N	N P I F D T	可	不可
<i>operand3</i>	S A M A U N P I F B* D T		可	可

\**operand3* のフォーマット B は、4 以下の長さでのみ使用できます。

構文要素の説明：

<b>GIVING</b>	結果フィールド：  GIVING 節を使用すると、 <i>operand2</i> は変更されず、 <i>operand3</i> に結果が入ります。
<i>operand1</i> FROM <i>operand2</i> <b>GIVING</b> <i>operand3</i>	オペランド：  <i>operand2</i> は被減数、 <i>operand1</i> は減数、 <i>operand3</i> は結果フィールドです。したがって、ステートメントは次と等しくなります。  < <i>oper3</i> > := < <i>oper2</i> > - < <i>oper1</i> >  オペランドのフォーマットについては、「フォーマット混合式のパフォーマンスについて」（『プログラミングガイド』）も参照してください。
<b>ROUNDED</b>	切り上げ：  キーワード <code>ROUNDED</code> を指定すると、結果は切り上げられます。切り上げについては、『プログラミングガイド』の「演算割り当てのルール」、「フィールドの切り捨てと切り上げ」を参照してください。

## 例

```

** Example 'SUBEX1': SUBTRACT
*****
DEFINE DATA LOCAL
1 #A (P2) INIT <50>
1 #B (P2)
1 #C (P1.1) INIT <2.4>
END-DEFINE
*
SUBTRACT 6 FROM #A
WRITE NOTITLE 'SUBTRACT 6 FROM #A          ' 10X '=' #A
*
SUBTRACT 6 FROM 11 GIVING #A
WRITE          'SUBTRACT 6 FROM 11 GIVING #A  ' 10X '=' #A
*
SUBTRACT 3 4 FROM #A GIVING #B
WRITE          'SUBTRACT 3 4 FROM #A GIVING #B ' 10X '=' #A '=' #B
*
SUBTRACT -3 -4 FROM #A GIVING #B
WRITE          'SUBTRACT -3 -4 FROM #A GIVING #B' 10X '=' #A '=' #B
*
SUBTRACT ROUNDED 2.06 FROM #C
WRITE          'SUBTRACT ROUNDED 2.06 FROM #C  ' 10X '=' #C
*
END

```

プログラム **SUBEX1** の出力：

```

SUBTRACT 6 FROM #A          #A: 44
SUBTRACT 6 FROM 11 GIVING #A #A: 5
SUBTRACT 3 4 FROM #A GIVING #B #A: 5 #B: -2
SUBTRACT -3 -4 FROM #A GIVING #B #A: 5 #B: 12
SUBTRACT ROUNDED 2.06 FROM #C #C: 0.3

```

# 126

## SUSPEND IDENTICAL SUPPRESS

---

■ 機能 .....	852
■ 構文説明 .....	852
■ 例 .....	852

SUSPEND IDENTICAL [SUPPRESS] [(*rep*)]

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT END OF PAGE](#) | [AT TOP OF PAGE](#) | [CLOSE PRINTER](#) | [DEFINE PRINTER](#) | [DISPLAY](#) | [EJECT](#) | [FORMAT](#) | [NEWPAGE](#) | [PRINT](#) | [SKIP](#) | [WRITE](#) | [WRITE TITLE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

---

SUSPEND IDENTICAL SUPPRESS ステートメントは、1レコードの処理に対して（同一フィールド値の出力を省略する）Natural セッションパラメータ設定 IS=ON を中止するために使用します。

セッションパラメータ IS（『[パラメータリファレンス](#)』）も参照してください。

## 構文説明

---

<i>(rep)</i>	<p>レポート指定：</p> <p>表記 (<i>rep</i>) を使用して、SUSPEND IDENTICAL SUPPRESS ステートメントを適用するレポートの ID を指定できます。</p> <p>範囲 0~31 の値、または <a href="#">DEFINE PRINTER</a> ステートメントを使用して割り当てた論理名を指定できます。</p> <p>(<i>rep</i>) 指定がない場合、SUSPEND IDENTICAL SUPPRESS ステートメントは最初のレポート（レポート 0）に適用されます。</p> <p>Natural で作成される出力レポートの形式を制御する方法については、『<a href="#">プログラミングガイド</a>』の「<a href="#">データ出力制御</a>」を参照してください。</p>
--------------	---

## 例

---

- [例 1 - SUSPEND IDENTICAL SUPPRESS を使用したプログラム](#)

- 例 2 - 前のプログラムと同様（ただし、SUSPEND IDENTICAL SUPPRESS は使用しない）

### 例 1 - SUSPEND IDENTICAL SUPPRESS を使用したプログラム

```

** Example 'SISEX1': SUSPEND IDENTICAL SUPPRESS
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 FIRST-NAME
  2 NAME
  2 CITY
1 VEH-VIEW VIEW OF VEHICLES
  2 PERSONNEL-ID
  2 MAKE
END-DEFINE
*
LIMIT 15
*
RD. READ EMPLOY-VIEW BY NAME STARTING FROM 'JONES'
/*
  SUSPEND IDENTICAL SUPPRESS
/*
  FD. FIND VEH-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (RD.)
  IF NO RECORDS FOUND
    MOVE '***NO CAR***' TO MAKE
  END-NOREC
  DISPLAY NOTITLE
    NAME (RD.) (IS=ON)
    FIRST-NAME (RD.) (IS=ON)
    MAKE (FD.)
  END-FIND
/*
END-READ
END

```

プログラム **SISEX1** の出力：

NAME	FIRST-NAME	MAKE
JONES	VIRGINIA	CHRYSLER
JONES	MARSHA	CHRYSLER
JONES	ROBERT	GENERAL MOTORS
JONES	LILLY	FORD
JONES	EDWARD	MG
JONES	EDWARD	GENERAL MOTORS

## SUSPEND IDENTICAL SUPPRESS

JONES	MARTHA	GENERAL MOTORS
JONES	LAUREL	GENERAL MOTORS
JONES	KEVIN	DATSUN
JONES	GREGORY	FORD
JONES	EDWARD	***NO CAR***
JOPER	MANFRED	***NO CAR***
JOUSSELIN	DANIEL	RENAULT
JUBE	GABRIEL	***NO CAR***
JUNG	ERNST	***NO CAR***
JUNKIN	JEREMY	***NO CAR***

### 例 2 - 前のプログラムと同様（ただし、SUSPEND IDENTICAL SUPPRESS は使用しない）

```
** Example 'SISEX2': SUSPEND IDENTICAL SUPPRESS (compare with SISEX1)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 FIRST-NAME
  2 NAME
  2 CITY
1 VEH-VIEW VIEW OF VEHICLES
  2 PERSONNEL-ID
  2 MAKE
END-DEFINE
*
LIMIT 15
RD. READ EMPLOY-VIEW BY NAME STARTING FROM 'JONES'
/*
/* SUSPEND IDENTICAL SUPPRESS      /* statement removed
/*
FD. FIND VEH-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (RD.)
  IF NO RECORDS FOUND
    MOVE '***NO CAR***' TO MAKE
  END-NOREC
  DISPLAY NOTITLE
    NAME (RD.) (IS=ON)
    FIRST-NAME (RD.) (IS=ON)
    MAKE (FD.)
  END-FIND
/*
END-READ
END
```



プログラム **SISEX2** の出力：

NAME	FIRST-NAME	MAKE
JONES	VIRGINIA	CHRYSLER
	MARSHA	CHRYSLER
		CHRYSLER
	ROBERT	GENERAL MOTORS
	LILLY	FORD
		MG
	EDWARD	GENERAL MOTORS
	MARTHA	GENERAL MOTORS
	LAUREL	GENERAL MOTORS
	KEVIN	DATSUN
JOPER	GREGORY	FORD
	EDWARD	***NO CAR***
JOUSSELIN	MANFRED	***NO CAR***
JUBE	DANIEL	RENAULT
JUNG	GABRIEL	***NO CAR***
JUNKIN	ERNST	***NO CAR***
	JEREMY	***NO CAR***



# 127      TERMINATE

---

▪ 機能 .....	858
▪ 構文説明 .....	858
▪ Natural 終了後に制御を受け取るプログラム .....	859
▪ 例 .....	859

TERMINATE [*operand1* [*operand2*]]

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 機能

TERMINATE ステートメントは、Natural セッションを終了するために使用します。TERMINATE ステートメントは、Natural プログラム内の任意の位置に指定できます。TERMINATE ステートメントが実行されると、ページ終了処理やループ終了処理は行われません。

TERMINATE ステートメントの動作は、STOP ステートメントの動作と一致しています。戻り値の処理はサポートされていません。

Natural RPC については、『[Natural リモートプロシージャコール \(RPC\)](#)』ドキュメントの「[サーバーに対する Natural ステートメントの注意事項](#)」を参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	N P I	可	不可
<i>operand2</i>	C S A	A U N P I F B D T L C	可	可

構文要素の説明：

<i>operand1</i>	<i>operand1</i> を使用して、Natural 終了時に制御を受け取るプログラムにリターンコードを渡します。例えば、リターンコード設定は、レジスタ 15 を介してオペレーティングシステムに渡すことができます。  <i>operand1</i> に指定できる値の範囲は 0~255 です。
<i>operand2</i>	<i>operand2</i> を使用して、終了後に制御を受け取るプログラムに追加情報を渡すことができます。

## Natural 終了後に制御を受け取るプログラム

Natural セッションの終了後、プロファイルパラメータ PROGRAM で指定された名前のプログラムに制御が渡ります。

### 例

```
** Example 'TEREX1': TERMINATE
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 SALARY (1)
*
1 #PNUM      (A8)
1 #PASSWORD (A8)
END-DEFINE
*
INPUT 'ENTER PASSWORD:' #PASSWORD
*
IF #PASSWORD NE 'USERPASS'
/*
  TERMINATE
/*
END-IF
*
INPUT 'ENTER PERSONNEL NUMBER:' #PNUM
*
FIND EMPLOY-VIEW WITH PERSONNEL-ID = #PNUM
  DISPLAY NAME SALARY (1)
END-FIND
*
END
```



# 128

## UPDATE

---

■ 機能 .....	862
■ 制限事項 .....	863
■ データベース固有の考慮事項 .....	863
■ 構文説明 .....	864
■ 例 .....	865

ストラクチャードモード構文

```
UPDATE [RECORD] [IN] [STATEMENT] [(r)]
```

レポートモード構文

```
UPDATE [RECORD] [IN] [STATEMENT] [(r)]
      [
          SET
          WITH
          USING
      ] { SAME [RECORD]
        { operand1=operand2 } ... }
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[ACCEPT/REJECT](#) | [AT BREAK](#) | [AT START OF DATA](#) | [AT END OF DATA](#) | [BACKOUT TRANSACTION](#) | [BEFORE BREAK PROCESSING](#) | [DELETE](#) | [END TRANSACTION](#) | [FIND](#) | [GET](#) | [GET SAME](#) | [GET TRANSACTION DATA](#) | [HISTOGRAM](#) | [LIMIT](#) | [PASSW](#) | [PERFORM BREAK PROCESSING](#) | [READ](#) | [RETRY](#) | [STORE](#)

関連機能グループ：「[データベースへのアクセスと更新](#)」

## 機能

UPDATE ステートメントは、データベースレコードの1つ以上のフィールドを更新するために使用します。更新するレコードは、[FIND](#)、[GET](#)、[READ](#)（または Adabas の場合は [STORE](#)）ステートメントで事前に選択されている必要があります。

### ホールド状態

UPDATE ステートメントの使用により、対応する [FIND](#) や [READ](#) ステートメントで選択された各レコードはホールド状態になります。

詳細については、「[レコードホールドロジック](#)（『[プログラミングガイド](#)』）」を参照してください。



## 制限事項

- UPDATE ステートメントを、更新するレコードを選択するステートメントと同一行に入力しないでください。
- UPDATE ステートメントは Entire System Server ビューには適用できません。

## データベース固有の考慮事項

<b>DL/I</b>	<p>UPDATE ステートメントは、DL/I データベースのセグメントを更新するために使用できます。必要に応じて、UPDATE ステートメントで指定された全フィールドを収容できるようにセグメント長は拡張されます。</p> <p>マルチプルバリュースフィールドまたはピリオディックグループが可変長として定義されている場合、UPDATE ステートメントに指定されたオカレンスのみがセグメントに書き込まれます。</p> <p>DL/I AIX フィールド名は、UPDATE ステートメントでは使用できません。ただし、AIX フィールドから成るソースフィールドを参照することにより、AIX フィールドを更新することができます。</p> <p>DL/I シーケンスフィールドは DL/I 制限のために更新できません。</p> <p>GSAM 制限により、UPDATE ステートメントは GSAM データベースには使用できません。</p>
<b>VSAM</b>	<p>VSAM プライマリキーは VSAM 制限のために更新できません。</p> <p>DL/I AIX フィールド名は、UPDATE ステートメントでは使用できません。ただし、AIX フィールドから成るソースフィールドを参照することにより、AIX フィールドを更新することができます。</p>
<b>SQL</b>	<p>UPDATE ステートメントは、データベーステーブルの行を更新するために使用できます。このステートメントは SQL の <code>UPDATE WHERE CURRENT OF CURSOR</code> (位置決め UPDATE) に相当し、最後に読み込まれた行のみを更新できます。</p> <p>プログラム内で変更された列 (フィールド) のみが、プログラム外 (マップの入力フィールドなど) で変更された可能性のある (実際に変更されている必要はない) 列に加えて更新されます。他のすべてのプラットフォームでは、すべての列が更新されます。</p> <p>ほとんどの SQL データベースで、<code>FIND SORTED BY</code> ステートメントや <code>READ LOGICAL</code> ステートメントによって読み込まれた行は更新できません。</p>

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A	A N P I F B D T L	不可	不可
<i>operand2</i>	C S A	A N P I F B D T L	可	不可

構文要素の説明：

(r)	<p>ステートメント参照：</p> <p>表記 (r) を使用して、更新するレコードが選択されたステートメントを示すことができます。r は、ソースコード行番号またはステートメントラベルとして指定できます。</p> <p>参照指定がなければ、UPDATE ステートメントでは、一番内側のアクティブな <b>READ</b> または <b>FIND</b> 処理ループが参照されます。アクティブな <b>READ</b> または <b>FIND</b> ループがない場合、このステートメントより前の最後の <b>GET</b> (または <b>STORE</b>) ステートメントが参照されます。</p> <p><b>注意:</b> UPDATE ステートメントは、参照している <b>READ</b> または <b>FIND</b> ループ内に指定する必要があります。</p>
<b>USING SAME</b>	<p>この節は、<b>DEFINE DATA</b> ステートメントを使用している場合は許可されません。UPDATE ステートメントは <b>DEFINE DATA</b> ステートメント内の定義に従って常にビュー全体を参照するからです。</p> <p>レコードバッファまたはフォーマットバッファのレイアウトは、<b>OBTAIN</b> ステートメントを使用して宣言できます。</p> <p><b>USING SAME</b> は、レポーティングモードで UPDATE ステートメントが参照するステートメントで読み込まれたフィールドと同じフィールドを更新に使用することを示します。この場合、各データベースフィールドに割り当てられた最新の値がフィールドの更新に使用されます。新しい値の割り当てがなければ、古い値が使用されます。</p> <p>更新するフィールドがマルチプルバリューフィールドやピリオディックグループの配列範囲で、この配列範囲に変数インデックスを使用する場合、最新の範囲が更新されます。つまり、レコードが読み込まれた後で、かつ UPDATE <b>USING SAME</b> (レポーティングモード) または UPDATE (ストラクチャードモード) の各ステートメントの実行前にインデックス変数が変更されると、更新される範囲は読み込み時の範囲と同じではなくなります。</p>
<b>SET/WITH</b> <i>operand1=operand2</i>	<p>この節は、レポーティングモードで更新するフィールドと値を指定するために使用します。</p>

この節は、DEFINE DATA ステートメントを使用している場合は許可されません。UPDATE ステートメントは DEFINE DATA ステートメント内の定義に従って常にビュー全体を参照するからです。

**注意:** DLI データベースの場合：SET/WITH 節を使用する場合、I/O（センシティブ）フィールドのみを指定できます。セグメントシーケンスフィールドは更新できません（代わりに DELETE および STORE を使用する必要があります）。

## 例

```

** Example 'UPDEX1S': UPDATE (structured mode)
**
** CAUTION: Executing this example will modify the database records!
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 CITY
*
1 #NAME (A20)
END-DEFINE
*
INPUT 'ENTER A NAME:' #NAME (AD=M)
IF #NAME = ' '
  STOP
END-IF
*
FIND EMPLOY-VIEW WITH NAME = #NAME
  IF NO RECORDS FOUND
    REINPUT WITH 'NO RECORDS FOUND' MARK 1
  END-NOREC
  INPUT 'NAME:      ' NAME (AD=0) /
        'FIRST NAME:' FIRST-NAME (AD=M) /
        'CITY:      ' CITY (AD=M)
  UPDATE
  END TRANSACTION
END-FIND
*
END

```

プログラム **SUBEX1S** の出力：

```
ENTER A NAME: BROWN
```

名前を入力して確認した後：

```
NAME: BROWN  
FIRST NAME: KENNETH  
CITY: DERBY
```

レポートモードの例はライブラリ SYSEXRM のプログラム **UPDEX1R** を参照してください。

# 129

## UPLOAD PC FILE

---

■ 機能 .....	869
■ 構文説明 .....	869
■ 例 .....	870

ストラクチャードモード構文

```

{ UPLOAD } { PC } [FILE] work-file-number [ONCE]
{ READ } { WORK }

{ RECORD operand1
  [AND] [SELECT] { [ { OFFSET n
                    FILLER nX } ]... operand2 }... }
[GIVING LENGTHoperand3]
[ AT [END] [OF] [FILE]
  statement ...
  END-ENDFILE ]
statement ...
END-WORK
    
```

レポートモード構文

```

{ UPLOAD } { PC } [FILE] work-file-number [ONCE]
{ READ } { WORK }

{ RECORD {operand1 [FILLER nX]} ...
  [AND] [SELECT] { [ OFFSET n
                    FILLER nX ]... operand2 }... }
[GIVING LENGTHoperand3]
[ AT [END] [OF] { statement
  [FILE] { DO statement ... DOEND } ] ]
statement ...
[LOOP]
    
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[CLOSE PC FILE](#) | [DOWNLOAD PC FILE](#) | [READ WORK FILE](#)

関連機能グループ：「[ワークファイル/PC ファイルの制御](#)」

## 機能

UPLOAD PC FILE ステートメントは、PC からメインフレーム プラットフォームにデータを転送するために使用します。

以下の項目も参照してください。

- 『*Natural Connection*』 および 『*Entire Connection*』 ドキュメント
- [READ WORK FILE](#) ステートメントの構文説明

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A G	A U N P I F B D T L C	可	可
<i>operand2</i>	S A G	A U N P I F B D T L C	可	可
<i>operand3</i>	S	I	可	可

フォーマット C は *Natural Connection* では無効です。

構文要素の説明：

<i>work-file-number</i>	使用するワークファイルの番号です。この番号は、 <i>Natural</i> に対する定義に従って、PC のワークファイル番号の 1 つに対応している必要があります。
<i>operand1-2</i>	フィールド指定：  <i>operand1</i> および <i>operand2</i> では、PC からアップロードするフィールドを指定します。データベースフィールドまたはユーザー定義変数を指定できます。
<i>statement</i>	UPLOAD PC FILE 処理で I/O ステートメントを指定することはできません。
<b>ONCE、SELECT、GIVING LENGTH RECORD</b>	オプション：  ONCE、SELECT、GIVING LENGTH オプションの詳細については、 <a href="#">READ WORK FILE</a> ステートメントの説明の中で対応する各セクションを参照してください。  PC ワークファイルには、RECORD オプションは使用できません。ランタイムに拒否されます。  データをアップロードするときにフィルタを定義する場合、標準のフィルタ表記の代わりにダミー変数を使用する必要があります。

<b>END-WORK</b>	UPLOAD PC FILE ステートメントを終了するには、Natural 予約キーワード END-WORK を使用する必要があります。
-----------------	--

## 例

次のプログラムでは、UPLOAD PC FILE ステートメントの使用方法を例示しています。データは最初に PC からアップロードされ、その後でメインフレームで処理されます。

```

** Example 'PCUPEX1': UPLOAD PC FILE
**
** NOTE: Example requires that Natural Connection is installed.
** CAUTION: Executing this example will modify the database records!
*****
DEFINE DATA LOCAL
01 EMPL VIEW OF EMPLOYEES
  02 PERSONNEL-ID
  02 INCOME
    03 SALARY (1)
*
01 #PID (A8) /* Personnel ID on PC
01 #NEW-INCREASE (N4) /* Increase for salary
END-DEFINE
*
UPLOAD PC FILE 7 #PID #NEW-INCREASE /* Data upload
*
  FIND EMPL WITH PERSONNEL-ID = #PID /* Data selection
  ADD #NEW-INCREASE TO SALARY (1) /* Data update on host
  UPDATE
  END TRANSACTION
  ESCAPE BOTTOM
END-FIND
*
END-WORK
END

```

プログラム **PCUPEX1** の出力：

プログラムを実行するとウィンドウが開くので、そこでアップロードするデータが含まれている PC ファイルの名前を指定します。その後、データが PC からアップロードされます。



# 130 WRITE

---

■ 機能 .....	872
■ 構文1 - 直接画面定義 .....	872
■ 構文1 - 説明 .....	873
■ 構文2 - フォーム／マップ使用 .....	881
■ 構文2 - 説明 .....	881
■ 例 .....	882

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT END OF PAGE](#) | [AT TOP OF PAGE](#) | [CLOSE PRINTER](#) | [DEFINE PRINTER](#) | [DISPLAY](#) | [EJECT](#) | [FORMAT](#) | [NEWPAGE](#) | [PRINT](#) | [SKIP](#) | [SUSPEND IDENTICAL SUPPRESS](#) | [WRITE TITLE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

WRITE ステートメントは、出力をフリーフォーマットで生成するために使用します。

WRITE ステートメントは、次の点で [DISPLAY](#) ステートメントと異なります。

- 行あふれをサポートします。レポートの幅が1行を超過する場合、あふれたフィールド（またはテキスト）は次の行に書き出されます。フィールドやテキストが行をまたがって出力されることはありません。
- デフォルトの列ヘッダーは出力されません。データの長さによって各フィールドの出力桁数が決まります。
- 配列の値やオカレンスは縦出力ではなく、横に並べられます。

『プログラミングガイド』の次のトピックも参照してください。

- [データ出力制御](#)
- [DISPLAY および WRITE ステートメント](#)
- [マルチプルバリューフィールドとピリオディックグループのインデックス表記](#)
- [DISPLAY VERT と WRITE ステートメントの例](#)
- [出力ページのレイアウト](#)

## 構文1 - 直接画面定義

```
WRITE [(rep)] [NOTITLE] [NOHDR]
  [(statement-parameters)]
  {
    [
      nX
      nT
      x/y
      T*field-name
      P*field-name
      ...
    ]
    {
      'text' [(attributes)]
      'c'(n) [(attributes)]
      [=] operand1 [(parameters)]
    }
  }
  ...
```

/

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 構文 1 - 説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A G N A U N P I F B D T L G O		可	不可

構文要素の説明：

( <i>rep</i> )	<p>レポート指定：</p> <p>プログラムで複数のレポートが生成される場合に、表記 (<i>rep</i>) を使用してレポートの ID を指定します。</p> <p>レポート ID として、範囲 0~31 の値、または <b>DEFINE PRINTER</b> ステートメントを使用して割り当てた論理名を指定できます。</p> <p>(<i>rep</i>) を指定しない場合、ステートメントは最初のレポート (レポート 0) に適用されます。</p> <p>このプリンタファイルが PC として Natural に定義されている場合、レポートは PC にダウンロードされます。<a href="#">例 5</a> を参照してください。</p> <p>Natural で作成される出力レポートの形式を制御する方法については、『プログラミングガイド』の「<a href="#">データ出力制御</a>」を参照してください。</p>
NOTITLE	<p>デフォルトページタイトルの省略：</p> <p>Natural は、WRITE ステートメントによって出力される各ページに 1 行のタイトル行を生成します。このタイトルには、ページ番号と日時が含まれます。時刻はプログラム実行の開始時に設定されます。このデフォルトのタイトル行は <b>WRITE TITLE</b> ステートメントを使用して上書きできます。また、WRITE ステートメントに NOTITLE オプションを指定することにより省略することもできます。</p> <p>例：</p>

	<ul style="list-style-type: none"> <li>■ デフォルトタイトルが生成されます。</li> </ul> <pre style="background-color: #f0f0f0; padding: 5px;">WRITE NAME</pre> <ul style="list-style-type: none"> <li>■ ユーザータイトルが生成されます。</li> </ul> <pre style="background-color: #f0f0f0; padding: 5px;">WRITE NAME WRITE TITLE 'user-title'</pre> <ul style="list-style-type: none"> <li>■ タイトルは生成されません。</li> </ul> <pre style="background-color: #f0f0f0; padding: 5px;">WRITE NOTITLE NAME</pre> <p><b>注意:</b></p> <ol style="list-style-type: none"> <li>1. NOTITLE オプションを使用すると、同じオブジェクト内でデータを同じレポートに書き込むすべての <b>DISPLAY</b>、<b>PRINT</b>、および <b>WRITE</b> ステートメントに適用されます。</li> <li>2. <b>WRITE</b> ステートメントの実行前にページオーバーフローがチェックされます。WRITE ステートメントの実行中にタイトルまたはトレイラ情報付きの新しいページが生成されることはありません。</li> </ol>
<b>NOHDR</b>	<p>列ヘッダーの省略：</p> <p>WRITE ステートメント自体では、列ヘッダーは生成されません。ただし、WRITE ステートメントを <b>DISPLAY</b> ステートメントと組み合わせて使用する場合、WRITE ステートメントの <b>NOHDR</b> オプションを使用して、DISPLAY ステートメントによって生成される列ヘッダーを省略できます。NOHDR オプションは、WRITE ステートメントの実行によって新しいページが出力される場合にのみ有効です。</p> <p>NOHDR オプションを使用しないと、この新しいページに DISPLAY ステートメントの列ヘッダー（存在する場合）が出力されます。NOHDR を使用すると、出力されません。</p>
<i>statement-parameters</i>	<p>ステートメントレベルでのパラメータ定義：</p> <p>1つまたは複数のパラメータをカッコで囲んで、ステートメントレベルで、つまり WRITE ステートメントの直後に指定できます。</p> <p>指定した各パラメータは、以前に <b>GLOBALS</b> コマンド、<b>SET GLOBALS</b> ステートメント（レポートモードでのみ）、または <b>FORMAT</b> ステートメントで指定した対応するパラメータを上書きします。</p> <p>複数のパラメータを指定する場合は、個々のパラメータを1つ以上の空白で区切る必要があります。各パラメータ指定を2行のステートメント行に分割することはできません。</p> <p><b>注意:</b> ここで適用されるパラメータ設定は、変数フィールドにのみ関連し、テキスト定数には影響しません。テキスト定数にフィールド属性を設定する</p>

	<p>場合は、この要素に属性を明示的に設定する必要があります（「<a href="#">要素（フィールド）レベルでのパラメータ定義</a>」を参照）。</p> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <a href="#">パラメータのリスト</a></li> <li>■ <a href="#">ステートメントおよび要素（フィールド）レベルでのパラメータ使用例</a></li> <li>■ <a href="#">例5 - ステートメント／要素（フィールド）レベルで '=' とパラメータを使用した WRITE ステートメント</a></li> </ul>
<p><i>nX, nT, x/y,</i> <i>T*field-name,</i> <i>P*field-name, '=', /,</i></p>	<p>フィールドの位置指定表記：</p> <p>「<a href="#">出力フォーマット定義</a>」セクションの「<a href="#">フィールドの位置指定表記</a>」を参照してください。</p>
<p><i>'text', 'c'(n), attributes,</i> <i>operand1, parameters</i></p>	<p>テキスト／属性割り当て：</p> <p>「<a href="#">出力フォーマット定義</a>」セクションの「<a href="#">テキスト／属性割り当て</a>」を参照してください。</p>

## パラメータのリスト

WRITE ステートメントで指定可能なパラメータ		指定 (S = ステートメントレベル、E = 要素レベル)
AD	属性定義	SE
AL	出力の英数字長	SE
BX	ボックス定義	SE
CD	カラー定義	SE
CV	制御変数	SE
DF	日付フォーマット	SE
DL	出力の表示長	SE
DY	ダイナミック属性	SE
EM	編集マスク	SE
FL	浮動小数点仮数長	SE
IS	重複抑制	SE
LS	行サイズ	S
MC	マルチプルバリューフィールド数	S
MP	レポートの最大ページ数	S
NL	出力の数値長	SE
PC	ピリオディックグループ数	S
PM	出力モード	SE

WRITE ステートメントで指定可能なパラメータ		指定 (S = ステートメントレベル、E = 要素レベル)
PS	ページサイズ*	S
SG	符号の位置	SE
UC	下線付き文字	S
ZP	ゼロ出力	SE

\* 配列のオカレンス数が PS 値を超えた場合は、NAT0303 エラーが出力されます。

各セッションパラメータの詳細については、『パラメータリファレンス』を参照してください。

『プログラミングガイド』の次のトピックも参照してください。

- 列ヘッダーの中央揃え - HC パラメータ
- 列ヘッダーの幅 - HW パラメータ
- ヘッダーの充填文字 - FC および GC パラメータ
- タイトルおよびヘッダーの下線付き文字 - UC パラメータ

## ステートメントおよび要素（フィールド）レベルでのパラメータ使用例

```

DEFINE DATA LOCAL
1 VARI (A4)      INIT <'1234'>          /*      Output
END-DEFINE      /*      Produced
*              /*      -----
WRITE           'Text'                 VARI          /*      Text 1234
WRITE (PM=I)   'Text'                 VARI          /*      Text 4321
WRITE         'Text' (PM=I)           VARI (PM=I)   /*      txeT 4321
WRITE         'Text' (PM=I)           VARI          /*      txeT 1234
END

```

「例5 - ステートメント／要素（フィールド）レベルで '=' とパラメータを使用した WRITE ステートメント」も参照してください。

## 出力フォーマット定義

$\left[ \begin{array}{l} nX \\ nT \\ x/y \\ T*field-name \\ P*field-name \\ / \end{array} \right] \dots$	$\left\{ \begin{array}{l} 'text' [(attributes)] \\ 'c' (n) [(attributes)] \\ ['=' ] operand1 [(parameters)] \end{array} \right\} \dots$
--	---

構文図で使用されている記号については、「構文記号」を参照してください。

## フィールドの位置指定表記

<b><i>nX</i></b>	<p>列の間隔：</p> <p>この表記により、列の間に <i>n</i> 個のスペースが挿入されます。 <i>n</i> をゼロにすることはできません。</p> <p>例：</p> <pre>WRITE NAME 5X SALARY</pre> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <a href="#">例2 - <i>nX</i>、<i>nT</i> 表記を使用した <b>WRITE</b> ステートメント</a> (後述)</li> <li>■ <a href="#">列の間隔 - SF パラメータと <i>nX</i> 表記</a> (『プログラミングガイド』)</li> </ul>
<b><i>nT</i></b>	<p>タブ設定：</p> <p><i>nT</i> 表記により、位置 <i>n</i> を出力するように位置指定 (タブ設定) されます。後方への位置指定はできません。</p> <p>次の例では、NAME は位置 25 から、SALARY は位置 50 から出力されます。</p> <pre>WRITE 25T NAME 50T SALARY</pre> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <a href="#">例2 - <i>nX</i>、<i>nT</i> 表記を使用した <b>WRITE</b> ステートメント</a> (後述)</li> <li>■ <a href="#">タブ設定 - <i>nT</i> 表記</a> (『プログラミングガイド』)</li> </ul>
<b><i>x/y</i></b>	<p><i>x/y</i> 位置指定：</p> <p><i>x/y</i> 表記により、次の要素は最後のステートメント出力の <i>x</i> 行下、列 <i>y</i> の先頭に配置されます。 <i>y</i> をゼロにすることはできません。同じ行で後方に位置指定することはできません。</p> <p>「<a href="#">位置指定表記 <i>x/y</i></a>」 (『プログラミングガイド』) も参照してください。</p>
<b><i>T*field-name</i></b>	<p>フィールドの相対位置指定：</p> <p>表記 <i>T*</i> では、前の <b>DISPLAY</b> ステートメントで使用したフィールドの指定出力桁に位置指定します。後方への位置指定はできません。</p> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <a href="#">例3 - <i>T*</i> 表記を使用した <b>WRITE</b> ステートメント</a> (後述)</li> <li>■ <a href="#">タブ表記 - <i>T*field</i></a> (『プログラミングガイド』)</li> </ul>
<b><i>P*field-name</i></b>	<p>フィールドおよび行の相対位置指定：</p>

	<p>表記 P* では、前の <b>DISPLAY</b> ステートメントで使用したフィールドの指定出力桁および行に位置指定します。縦出力モードと組み合わせてよく使用します。後方への位置指定はできません。</p> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <b>例 4 - P* 表記を使用した WRITE ステートメント</b> (後述)</li> <li>■ <b>タブ表記 P*field</b> (『プログラミングガイド』)</li> </ul>
'='	<p>フィールドヘッダーの後に位置するフィールド内容：</p> <p>フィールドの前に配置した場合は、等号 '=' により、<b>DEFINE DATA</b> ステートメントまたは DDM の定義に従って、フィールドヘッダーはフィールド内容の前に表示されます。</p> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <b>例 1 - '='、'text'、'/' を使用した WRITE ステートメント</b></li> <li>■ <b>例 5 - ステートメント／要素レベルで '=' とパラメータを使用した WRITE ステートメント</b></li> </ul>
/	<p>行送り - スラッシュ表記：</p> <p>フィールドまたはテキスト要素の間に配置した場合は、スラッシュ (/) により次の出力行の先頭に位置指定されます。</p> <p>例：</p> <pre>WRITE NAME / SALARY</pre> <p>複数の行送りをするために、複数のスラッシュ (/) 表記を使用できます。</p> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <b>例 1 - '='、'text'、'/' を使用した WRITE ステートメント</b> (後述)</li> <li>■ <b>行送り - スラッシュ表記</b> (『プログラミングガイド』)</li> <li>■ <b>例 2 - WRITE ステートメントでの行送り</b> (『プログラミングガイド』)</li> </ul>

### テキスト／属性割り当て

'text'	<p>テキスト割り当て：</p> <p>一重引用符で囲まれた文字列が表示されます。</p>
--------	---



	<p>例：</p> <pre>WRITE 'EMPLOYEE' NAME 'MARITAL/STATUS' MAR-STAT</pre> <p>以下の項目も参照してください。</p> <ul style="list-style-type: none"> <li>■ <a href="#">例1 - '='、'text'、'/' を使用した WRITE ステートメント</a> (後述)</li> <li>■ 「テキスト表記」、 「ステートメントで使用するテキストの定義」 (『プログラミングガイド』)</li> </ul>
<i>'c'(n)</i>	<p>文字の繰り返し：</p> <p>フィールド値の直前に、一重引用符で囲まれた文字列が <i>n</i> 回表示されます。</p> <p>例：</p> <pre>WRITE '*' (5) '=' NAME</pre> <p>結果</p> <pre>***** SMITH</pre> <p>「テキスト表記」の「フィールド値の前に <i>n</i> 回表示される文字の定義」 (『プログラミングガイド』) も参照してください。</p>
<i>attributes</i>	<p>フィールド表現と色属性：</p> <p>テキスト／フィールド表示にさまざまな属性を割り当てることができます。これらの属性と構文については、以下の「出力属性」で説明します。</p> <p>例：</p> <pre>WRITE 'TEXT' (BGR) WRITE 'TEXT' (B) WRITE 'TEXT' (BBLC)</pre>
<i>operand1</i>	<p>書き込まれるフィールド：</p> <p><i>operand1</i> では、内容がこの位置に書き込まれるフィールドを指定します。</p> <p>実行時にオカレンス数を変更できる、範囲による配列は指定できません。</p> <p><b>注意：</b> DL/I データベースの場合：DL/I AIX フィールドは、パラメータ PROCSEQ で指定した AIX とともに PCB を使用している場合にのみ表示できます。それ以外の場合は、ランタイムに Natural によってエラーメッセージが返されます。</p>
<i>parameters</i>	<p>要素 (フィールド) レベルでのパラメータ定義：</p> <p>1 つまたは複数のパラメータをカッコで囲んで、要素 (フィールド) レベルで、つまり <i>operand1</i> の直後に指定できます。この方法で指定した各パラメータは、以前に <b>ステート</b></p>

メントレベル、GLOBALS コマンド、SET GLOBALS ステートメント（レポートモードでのみ）、または FORMAT ステートメントで指定した対応するパラメータを上書きします。

複数のパラメータを指定する場合は、各エントリ間に1つ以上の空白を配置する必要があります。エントリを2行のステートメント行に分割することはできません。

以下の項目も参照してください。

- [パラメータのリスト](#)
- [ステートメントおよび要素（フィールド）レベルでのパラメータ使用例](#)

## 出力属性

*attributes* は、テキスト表示に使用される出力属性を示します。可能な属性：

```

{
  {
    AD=AD-value ...
    BX=BX-value ...
    CD=CD-value ...
    PM=PM-value ...
  } ...
  {
    AD-value ...
    CD-value ...
  } ...
}

```

指定可能なセッションパラメータ値については、『パラメータリファレンス』ドキュメントの該当するセクションを参照してください。

- 「AD - 属性定義」の「フィールド表現」
- CD - カラー定義
- BX - 外枠の定義
- PM - 出力モード



**注意:** コンパイラは、実際には1つの出力フィールドに複数の属性値を受け入れます。例えば、「AD=BDI」と指定できます。ただし、この場合は最後の値のみが適用されます。示した例では、値 I のみが有効になり、出力フィールドは強調表示されます。

## 構文 2 - フォーム／マップ使用

```
WRITE [(rep)] [NOTITLE] [NOHDR] [USING] { FORM MAP } operand1 [operand2 ...]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 構文 2 - 説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S	A	不可	不可
<i>operand2</i>	S A G N	A U N P I F B D T L	可	不可

構文要素の説明：

<b>[USING] FORM[USING] MAP</b>	<p>定義済みフォーム／マップレイアウトの使用</p> <p>このオプションでは、あらかじめNatural マップエディタで定義しておいたフォーム／マップレイアウトを使用するように指定できます。</p> <p>WRITE ステートメントで使用するマップレイアウトでは、各マップ出力ごとの自動的なページ換えは行われません。</p> <p>LS パラメータはマップで定義した LS 設定よりも 1 バイト大きく指定する必要があります。</p>
<i>operand1</i>	<p><b>Form/Map 名：</b></p> <p><i>operand1</i> は、使用するフォーム／マップの名前です。</p>
<i>operand2</i>	<p><b>書き込まれるフィールド：</b></p> <p><i>operand2</i> は、出力するフィールドの名前です。</p> <p><i>operand1</i> が定数で、<i>operand2</i> が省略されている場合、コンパイル時にマップソースからフィールドが取り込まれます。</p> <p>フィールドは、番号、シーケンス、フォーマット、長さ、オカレンス数（配列の場合）について、参照されているフォーム／マップに準拠している必要があります。そうでない場合、エラーが発生します。</p>
<b>NOTITLE/NOHDR</b>	<p>タイトル行／列ヘッダーの省略：</p>

NOTITLE および NOHDR については、WRITE ステートメントの構文1を参照してください。

## 例

- 例 1 - '='、'text'、'/' を使用した WRITE ステートメント
- 例 2 - nX、nT 表記を使用した WRITE ステートメント
- 例 3 - T\* 表記を使用した WRITE ステートメント
- 例 4 - P\* 表記を使用した WRITE ステートメント
- 例 5 - ステートメント／要素（フィールド）レベルで '=' とパラメータを使用した WRITE ステートメント
- 例 6 - Natural に PC として定義された出力ファイルを使用したレポート指定

### 例 1 - '='、'text'、'/' を使用した WRITE ステートメント

```
** Example 'WRTEX1': WRITE (with '=', 'text', '/')
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 FULL-NAME
    3 FIRST-NAME
    3 MIDDLE-I
    3 NAME
  2 CITY
  2 COUNTRY
END-DEFINE
*
LIMIT 1
READ EMPL-VIEW BY NAME
/*
  WRITE NOTITLE
    '=' NAME '=' FIRST-NAME '=' MIDDLE-I //
    'L O C A T I O N' /
    'CITY: ' CITY /
    'COUNTRY:' COUNTRY //
/*
END-READ
END
```

プログラム **WRTEX1** の出力：

```
NAME: ABELLAN                FIRST-NAME: KEPA                MIDDLE-I:
L O C A T I O N
CITY:   MADRID
COUNTRY: E
```

## 例 2 - nX、nT 表記を使用した WRITE ステートメント

```
** Example 'WRTEX2': WRITE (with nX, nT notation)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 JOB-TITLE
END-DEFINE
*
LIMIT 4
READ EMPL-VIEW BY NAME
  WRITE NOTITLE 5X NAME 50T JOB-TITLE
END-READ
END
```

プログラム **WRTEX2** の出力：

```
ABELLAN                MAQUINISTA
ACHIESON               DATA BASE ADMINISTRATOR
ADAM                   CHEF DE SERVICE
ADKINSON               PROGRAMMER
```

## 例 3 - T\* 表記を使用した WRITE ステートメント

```
** Example 'WRTEX3': WRITE (with T* notation)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CITY
  2 SALARY (1)
END-DEFINE
*
LIMIT 5
READ EMPL-VIEW BY CITY STARTING FROM 'ALBU'
  DISPLAY NOTITLE CITY NAME SALARY (1)
  AT BREAK CITY
  /*
  WRITE / 'CITY AVERAGE:' T*SALARY (1) AVER(SALARY(1)) //
```

## WRITE

---

```
 /*
  END-BREAK
  END-READ
  END
```

プログラム **WRTEX3** の出力：

CITY	NAME	ANNUAL SALARY
ALBUQUERQUE	HAMMOND	22000
ALBUQUERQUE	ROLLING	34000
ALBUQUERQUE	FREEMAN	34000
ALBUQUERQUE	LINCOLN	41000
CITY AVERAGE:		32750
ALFRETON	GOLDBERG	4800
CITY AVERAGE:		4800

### 例 4 - P\* 表記を使用した WRITE ステートメント

```
** Example 'WRTEX4': WRITE (with P* notation)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 CITY
  2 BIRTH
  2 SALARY (1)
END-DEFINE
*
LIMIT 3
READ EMPL-VIEW BY CITY FROM 'N'
  DISPLAY NOTITLE NAME CITY
    VERT AS 'BIRTH/SALARY' BIRTH (EM=YYYY-MM-DD) SALARY (1)
  SKIP 1
  AT BREAK CITY
    WRITE / 'CITY AVERAGE' P*SALARY (1) AVER(SALARY (1)) //
  END-BREAK
END-READ
END
```

プログラム **WRTEX4** の出力：

NAME	CITY	BIRTH SALARY
-----	-----	-----
WILCOX	NASHVILLE	1970-01-01 38000
MORRISON	NASHVILLE	1949-07-10 36000
CITY AVERAGE		37000
BOYER	NEMOURS	1955-11-23 195900
CITY AVERAGE		195900

**例 5- ステートメント／要素（フィールド）レベルで '=' とパラメータを使用した WRITE ステートメント**

```

** Example 'WRTEX5': WRITE (using '=', statement/element parameters)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 PERSONNEL-ID
  2 PHONE
END-DEFINE
*
LIMIT 2
READ EMPL-VIEW BY NAME
  WRITE NOTITLE (AL=16 NL=8)
    '=' PERSONNEL-ID '=' NAME '=' PHONE (AL=10 EM=XXX-XXXXXXX)
END-READ
END

```

プログラム **WRTEX5** の出力：

PERSONNEL ID: 60008339	NAME: ABELLAN	TELEPHONE: 435-6726
PERSONNEL ID: 30000231	NAME: ACHIESON	TELEPHONE: 523-341

### 例 6 - Natural に PC として定義された出力ファイルを使用したレポート指定

```
** Example 'PCDIEX1': DISPLAY and WRITE to PC
**
** NOTE: Example requires that Natural Connection is installed.
*****
DEFINE DATA LOCAL
01 PERS VIEW OF EMPLOYEES
  02 PERSONNEL-ID
  02 NAME
  02 CITY
END-DEFINE
*
FIND PERS WITH CITY = 'NEW YORK'          /* Data selection
  WRITE (7) TITLE LEFT 'List of employees in New York' /
  DISPLAY (7)          /* (7) designates the output file (here the PC).
    'Location'  CITY
    'Surname'   NAME
    'ID'        PERSONNEL-ID
END-FIND
END
```



# 131

## WRITE TITLE

---

▪ 機能 .....	888
▪ 制限事項 .....	889
▪ 構文説明 .....	889
▪ 例 .....	892

```
WRITE [(rep)]TITLE [LEFT [JUSTIFIED]] [UNDERLINED]
  [(statement-parameters)]
  { [ [ nX ] ] { 'text' [(attributes)] } }
  { [ [ nT ] ] { 'c'(n) [(attributes)] } }
  { [ [ x/y ] ] ... { [=] operand1 [(parameters)] } ... }
  [SKIP operand2 [LINES]]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT END OF PAGE](#) | [AT TOP OF PAGE](#) | [CLOSE PRINTER](#) | [DEFINE PRINTER](#) | [DISPLAY](#) | [EJECT](#) | [FORMAT](#) | [NEWPAGE](#) | [PRINT](#) | [SKIP](#) | [SUSPEND](#) | [IDENTICAL SUPPRESS](#) | [WRITE](#) | [WRITE TRAILER](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

WRITE TITLE ステートメントは、デフォルトのページタイトルではなくユーザー指定のページタイトルを表示するために使用します。このステートメントは、新しいページが始まるたびに実行されます。

『プログラミングガイド』の次のセクションも参照してください。

- データ出力制御
- レポート指定 - (rep) 表記
- 出力ページのレイアウト
- ページタイトル、改ページ、空行
- 独自ページタイトル定義 - WRITE TITLE ステートメント
- テキスト表記

## 処理

このステートメントは非手続き型なので、プログラム内の位置ではなくイベントによって実行されます。

レポートが、異なるオブジェクト内のステートメントによって作成される場合、WRITE TITLE ステートメントは、新ページを作成するステートメントと同じオブジェクトで使用されている場合に限って実行されます。

## 制限事項

- WRITE TITLE は 1 レポートにつき 1 回だけ指定できます。
- WRITE TITLE は特殊条件ステートメントブロック内では指定できません。
- WRITE TITLE はサブルーチン内では指定できません。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A G N A U N P I F B D T L G O		可	不可
<i>operand2</i>	C S	N P I B	可	不可

構文要素の説明：

<i>(rep)</i>	<p>レポート指定：</p> <p>複数のレポートを作成するときは、表記 (<i>rep</i>) を使用して、WRITE TITLE ステートメントを適用するレポートの ID を指定できます。</p> <p>レポート ID として、範囲 0~31 の値、または DEFINE PRINTER ステートメントを使用して割り当てた論理名を指定できます。</p> <p>(<i>rep</i>) 指定がない場合、WRITE TITLE ステートメントは最初のレポート (レポート 0) に適用されます。</p> <p>Natural で作成される出力レポートの形式を制御する方法については、『プログラミングガイド』の「データ出力制御」を参照してください。</p>
LEFT JUSTIFIED UNDERLINED	<p>桁揃えおよび下線</p> <p>デフォルトでは、ページタイトルは中央揃えで下線なしです。LEFT JUSTIFIED と UNDERLINED を指定して上書きできます。</p> <p>UNDERLINED を指定した場合、下線文字 (デフォルトの文字または FORMAT ステートメントのセッションパラメータ UC で指定された文字) が、タイトルの下に行サイズ (セッションパラメータ LS を参照) の幅だけ出力されます。</p> <p>Natural では、すべての間隔およびタブ指定を適用して行を作成してから、行全体を中央揃えします。例えば、最初の要素としての 10T 表記により、中央揃えされたヘッダーは右側の 5 の位置に配置されます。</p>

<p><i>statement-parameters</i></p>	<p>ステートメントレベルでのパラメータ定義：</p> <p>1つまたは複数のパラメータをカッコで囲んで、ステートメントレベルで、つまり WRITE TITLE ステートメントの直後に指定できます。この方法で指定した各パラメータは、以前に GLOBALS コマンド、<a href="#">SET GLOBALS</a> ステートメント（レポートモードでのみ）、または <a href="#">FORMAT</a> ステートメントで指定したパラメータを上書きします。</p> <p>複数のパラメータを指定する場合は、各エントリ間に1つ以上の空白を配置する必要があります。エントリを2行のステートメント行に分割することはできません。</p> <p><b>注意：</b>ここで適用されるパラメータ設定は、変数フィールドにのみ関連し、テキスト定数には影響しません。テキスト定数にフィールド属性を設定する場合は、この要素に属性を明示的に設定する必要があります（「<a href="#">要素（フィールド）レベルでのパラメータ定義</a>」を参照）。</p> <p>使用可能なパラメータについては、『ステートメント』ドキュメントの「WRITE」で「<a href="#">パラメータのリスト</a>」を参照してください。</p>
<p><i>nX</i></p> <p><i>nT</i></p> <p><i>x/y</i></p>	<p>フォーマット表記と要素間隔：</p> <p>「<a href="#">フォーマット表記と要素間隔</a>」（後述）を参照してください。</p>
<p>'<i>text</i>'</p> <p>'<i>c</i>' (<i>n</i>)</p> <p><i>attributes</i></p>	<p>テキスト／属性割り当て：</p> <p>「<a href="#">テキスト／属性割り当て</a>」（後述）を参照してください。</p>
<p><i>operand1</i></p>	<p>タイトル内に表示するフィールド：</p> <p><i>operand1</i> は、タイトル内に表示するフィールド（複数可）を表します。</p> <p>実行時にオカレンス数を変更できる、範囲による配列は指定できません。</p>
<p><i>parameters</i></p>	<p>要素（フィールド）レベルでのパラメータ定義：</p> <p>1つまたは複数のパラメータをカッコで囲んで、要素（フィールド）レベルで、つまり <i>operand1</i> の直後に指定できます。この方法で指定した各パラメータは、以前に <a href="#">ステートメントレベル</a>、GLOBALS コマンド、<a href="#">SET GLOBALS</a> ステートメント（レポートモードでのみ）、または <a href="#">FORMAT</a> ステートメントで指定した対応するパラメータを上書きします。</p> <p>複数のパラメータを指定する場合は、各エントリ間に1つ以上の空白を配置する必要があります。エントリを2行のステートメント行に分割することはできません。</p> <p>使用可能なパラメータについては、『ステートメント』ドキュメントの「WRITE」で「<a href="#">パラメータのリスト</a>」を参照してください。</p>

<b>SKIP operand2 LINES</b>	<p>省略される行：</p> <p>SKIPを使用して、タイトル行の下に何行か空けることができます。空ける行数は数値定数または数値変数の内容で指定します。</p> <p><b>注意:</b> WRITE TITLE の後の SKIP は独立したステートメントでなく、常に WRITE TITLE ステートメントの SKIP 節として解釈されます。WRITE TITLE ステートメントの後に独立した SKIP ステートメントが必要な場合は、セミコロン (;) を使用して2つのステートメントを互いに区別します。</p>
----------------------------	---

### フォーマット表記と要素間隔


<b>nX</b>	<p>列の間隔：</p> <p>この表記により、列の間に <math>n</math> 個のスペースが挿入されます。 <math>n</math> をゼロにすることはできません。</p>
<b>nT</b>	<p>タブ設定：</p> <p><math>nT</math> 表記により、位置 <math>n</math> を出力するように位置指定（タブ設定）されます。後方への位置指定はできません。</p>
<b>x/y</b>	<p><math>x/y</math> 位置指定：</p> <p>次の要素は最後のステートメント出力の <math>x</math> 行下、列 <math>y</math> の先頭に配置されます。 <math>y</math> をゼロにすることはできません。同じ行で後方に位置指定することはできません。</p>

### テキスト／属性割り当て

<b>'text'</b>	<p>テキスト割り当て：</p> <p>一重引用符で囲まれた文字列が表示されます。</p>
<b>'c'(n)</b>	<p>文字の繰り返し：</p> <p>フィールド値の直前に、一重引用符で囲まれた文字列が <math>n</math> 回表示されます。</p>
<b>attributes</b>	<p>フィールド表現と色属性：</p> <p>テキスト／フィールド表示にさまざまな属性を割り当てることができます。これらの属性と構文については、以下の「出力属性」で説明します。</p> <p>例：</p> <pre>WRITE TITLE 'TEXT' (BGR) WRITE TITLE 'TEXT' (B) WRITE TITLE 'TEXT' (BBLC)</pre>

## 出力属性

*attributes* は、テキスト表示に使用される出力属性を示します。可能な属性：



指定可能なセッションパラメータ値については、『パラメータリファレンス』ドキュメントの該当するセクションを参照してください。

- 「AD - 属性定義」の「フィールド表現」
- CD - カラー定義
- BX - 外枠の定義
- PM - 出力モード



**注意:** コンパイラは、実際には1つの出力フィールドに複数の属性値を受け入れます。例えば、「AD=BDI」と指定できます。ただし、この場合は最後の値のみが適用されます。示した例では、値 I のみが有効になり、出力フィールドは強調表示されます。

## 例

```
** Example 'WTIEX1': WRITE (with TITLE option)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 CITY
  2 JOB-TITLE
END-DEFINE
*
*
FORMAT LS=70
*
WRITE TITLE LEFT JUSTIFIED UNDERLINED
      *TIME 3X 'PEOPLE LIVING IN NEW YORK CITY'
      11X 'PAGE:' *PAGE-NUMBER
SKIP 1
*
FIND EMPL-VIEW WITH CITY = 'NEW YORK'
```

```
DISPLAY NAME FIRST-NAME 3X JOB-TITLE  
END-FIND  
END
```

プログラム **WTIEX1** の出力：

```
09:33:16.5  PEOPLE LIVING IN NEW YORK CITY  PAGE: 1
```

```
-----  
          NAME                FIRST-NAME                CURRENT  
                                POSITION  
-----
```

```
RUBIN                SYLVIA                SECRETARY  
WALLACE              MARY                 ANALYST
```





# 132 WRITE TRAILER

---

▪ 機能 .....	896
▪ 制約 .....	897
▪ 構文説明 .....	897
▪ 例 .....	900

```
WRITE [(rep)] [TRAILER LEFT [JUSTIFIED]] [UNDERLINED]
  [(statement-parameters)]
  {
    [
      nX
      nT
      x/y
    ] ... {
      'text' [(attributes)]
      'c'(n) [(attributes)]
      [=] operand1 [(parameters)]
    } ...
  }
  [SKIP operand2 [LINES]]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[AT END OF PAGE](#) | [AT TOP OF PAGE](#) | [CLOSE PRINTER](#) | [DEFINE PRINTER](#) | [DISPLAY](#) | [EJECT](#) | [FORMAT](#) | [NEWPAGE](#) | [PRINT](#) | [SKIP](#) | [SUSPEND IDENTICAL SUPPRESS](#) | [WRITE](#) | [WRITE TITLE](#)

関連機能グループ：「[出力レポートの作成](#)」

## 機能

WRITE TRAILER ステートメントは、ページの下にテキストまたは変数の内容を入力するために使用します。

『プログラミングガイド』の次のセクションも参照してください。

- データ出力制御
- レポート指定 - (rep) 表記
- 出力ページのレイアウト
- ページトレーラ - WRITE TRAILER ステートメント
- テキスト表記

## 処理

このステートメントは非手続き型なので、プログラム内の位置ではなくイベントによって実行されます。

このステートメントは、エンドオブページ条件またはエンドオブデータ条件が検出されたとき、あるいは SKIP や NEWPAGE ステートメントによって改ページが行われたときに実行されます。EJECT ステートメントでは実行されません。

エンドオブページ条件は、DISPLAY/WRITE ステートメントの処理が完了してから評価されます。DISPLAY/WRITE ステートメントで複数行の出力を行う場合、エンドオブページ条件になる前に物理ページのオーバーフローを起こすことがあります。

レポートが、異なるオブジェクト内のステートメントによって作成される場合、WRITE TRAILER ステートメントは、END-OF-PAGE（ページ終了）処理を行うステートメントと同じオブジェクトで使用されている場合に限って実行されます。

## 論理ページサイズ

トレーラ情報を正しく同一ページの下に出力するには、論理ページサイズ（セッションパラメータ PS で指定）を物理ページサイズより小さく指定します。

## 制約

- WRITE TRAILER は 1 レポートにつき 1 回だけ指定できます。
- WRITE TRAILER は特殊条件ステートメントブロック内では指定できません。
- WRITE TRAILER はサブルーチン内では指定できません。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	S A G N	A U N P I F B D T L G O	可	不可
<i>operand2</i>	C S	N P I B	可	不可

構文要素の説明：

<i>(rep)</i>	<p>レポート指定：</p> <p>複数のレポートを作成するときは、表記 (<i>rep</i>) を使用して、WRITE TRAILER ステートメントを適用するレポートの ID を指定できます。</p> <p>レポート ID として、範囲 0~31 の値、または <b>DEFINE PRINTER</b> ステートメントを使用して割り当てた論理名を指定できます。</p> <p>(<i>rep</i>) 指定がない場合、WRITE TRAILER ステートメントは最初のレポート（レポート 0）に適用されます。</p> <p>Natural で作成される出力レポートの形式を制御する方法については、『プログラミングガイド』の「データ出力制御」を参照してください。</p>
<b>LEFT JUSTIFIED</b> <b>UNDERLINED</b>	<p>桁揃えおよび下線</p> <p>デフォルトでは、トレーラ行は中央揃えで下線なしです。</p>

	<p>LEFT JUSTIFIED と UNDERLINED を指定して上書きできます。</p> <p>UNDERLINEDを指定した場合、下線文字（デフォルトの文字またはセッションパラメータ UC で指定された文字）が、トレイラの下に行サイズ（セッションパラメータ LS）の幅だけ出力されます。</p> <p>Naturalでは、すべての間隔およびタブ指定を適用して行を作成してから、行全体を中央揃えします。例えば、最初の要素としての 10T 表記により、中央揃えされたヘッダーは右側の 5 の位置に配置されます。</p>
<i>statement-parameters</i>	<p>ステートメントレベルでのパラメータ定義：</p> <p>1つまたは複数のパラメータをカッコで囲んで、ステートメントレベルで、つまり WRITE TRAILER ステートメントの直後に指定できます。この方法で指定した各パラメータは、以前に GLOBALS コマンド、<a href="#">SET GLOBALS</a> ステートメント（レポートモードでのみ）、または <a href="#">FORMAT</a> ステートメントで指定したパラメータを上書きします。</p> <p>複数のパラメータを指定する場合は、各エントリ間に1つ以上の空白を配置する必要があります。エントリを2行のステートメント行に分割することはできません。</p> <p><b>注意:</b> ここで適用されるパラメータ設定は、変数フィールドにのみ関連し、テキスト定数には影響しません。テキスト定数にフィールド属性を設定する場合は、この要素に属性を明示的に設定する必要があります（「<a href="#">要素（フィールド）レベルでのパラメータ定義</a>」を参照）。</p> <p>使用可能なパラメータについては、『ステートメント』ドキュメントの「WRITE」で「<a href="#">パラメータのリスト</a>」を参照してください。</p>
<i>nX</i> <i>nT</i> <i>x/y</i>	<p>フォーマット表記と要素間隔：</p> <p>「<a href="#">フォーマット表記と要素間隔</a>」（後述）を参照してください。</p>
' <i>text</i> ' ' <i>c</i> ( <i>n</i> )' <i>attributes</i>	<p>テキスト／属性割り当て：</p> <p>「<a href="#">テキスト／属性割り当て</a>」（後述）を参照してください。</p>
<i>operand1</i>	<p>トレイラ情報：</p> <p><i>operand1</i> は、トレイラ情報として出力するフィールド（複数可）を表します。</p> <p>実行時にオカレンス数を変更できる、範囲による配列は指定できません。</p>
<i>parameters</i>	<p>要素（フィールド）レベルでのパラメータ定義：</p> <p>1つまたは複数のパラメータをカッコで囲んで、要素（フィールド）レベルで、つまり <i>operand1</i> の直後に指定できます。この方法で指定した各パラメータは、以前に<a href="#">ステートメントレベル</a>、GLOBALS コマンド、<a href="#">SET GLOBALS</a> ス</p>

	<p>テートメント（レポーティングモードでのみ）、または <b>FORMAT</b> ステートメントで指定した対応するパラメータを上書きします。</p> <p>複数のパラメータを指定する場合は、各エントリ間に1つ以上の空白を配置する必要があります。エントリを2行のステートメント行に分割することはできません。</p> <p>使用可能なパラメータについては、『ステートメント』ドキュメントの「WRITE」で「<a href="#">パラメータのリスト</a>」を参照してください。</p>
<b>SKIP operand2 LINES</b>	<p>省略される行：</p> <p>SKIPを使用して、トレイラ行の下に何行か空けることができます。空ける行数（<i>operand2</i>）は数値定数または数値変数の内容で指定します。</p> <p><b>注意:</b> WRITE TRAILER の後の SKIP は独立したステートメントでなく、常に WRITE TRAILER ステートメントの SKIP 節として解釈されます。WRITE TRAILER ステートメントの後に独立した <b>SKIP</b> ステートメントが必要な場合は、セミコロン (;) を使用して2つのステートメントを互いに区別します。</p>

### フォーマット表記と要素間隔

<i>nX</i>	<p>列の間隔：</p> <p>この表記により、列の間に <i>n</i> 個のスペースが挿入されます。 <i>n</i> をゼロにすることはできません。</p>
<i>nT</i>	<p>タブ設定：</p> <p><i>nT</i> 表記により、位置 <i>n</i> を出力するように位置指定（タブ設定）されます。後方への位置指定はできません。</p>
<i>x/y</i>	<p><i>x/y</i> 位置指定：</p> <p>次の要素は最後のステートメント出力の <i>x</i> 行下、列 <i>y</i> の先頭に配置されます。 <i>y</i> をゼロにすることはできません。同じ行で後方に位置指定することはできません。</p>


### テキスト／属性割り当て

'text'	<p>テキスト割り当て：</p> <p>一重引用符で囲まれた文字列が表示されます。</p>
'c'(n)	<p>文字の繰り返し：</p> <p>フィールド値の直前に、一重引用符で囲まれた文字列が <i>n</i> 回表示されます。</p>
attributes	<p>フィールド表現と色属性：</p> <p>テキスト／フィールド表示にさまざまな属性を割り当てることができます。これらの属性と構文については、以下の「<a href="#">出力属性</a>」で説明します。</p>

	例：  WRITE TRAILER 'TEXT' (BGR) WRITE TRAILER 'TEXT' (B) WRITE TRAILER 'TEXT' (BBLC)
--	---

出力属性

*attributes* は、テキスト表示に使用される出力属性を示します。可能な属性：



指定可能なセッションパラメータ値については、『パラメータリファレンス』ドキュメントの該当するセクションを参照してください。

- 「AD - 属性定義」の「フィールド表現」
- CD - カラー定義
- BX - 外枠の定義
- PM - 出力モード

**注意:** コンパイラは、実際には1つの出力フィールドに複数の属性値を受け入れます。例えば、「AD=BDI」と指定できます。ただし、この場合は最後の値のみが適用されます。示した例では、値 1 のみが有効になり、出力フィールドは強調表示されます。

例

```

** Example 'WTLEX1': WRITE (with TRAILER option)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 CITY
  2 JOB-TITLE
END-DEFINE
*
FORMAT PS=15
WRITE TITLE LEFT JUSTIFIED UNDERLINED
    
```

```

*TIME 3X 'PEOPLE LIVING IN BARCELONA'
14X 'PAGE:' *PAGE-NUMBER
SKIP 1
*
WRITE TRAILER LEFT JUSTIFIED UNDERLINED
/ 'CITY OF BARCELONA REGISTER'
*
LIMIT 10
FIND EMPL-VIEW WITH CITY = 'BARCELONA'
  DISPLAY NAME FIRST-NAME 3X JOB-TITLE
END-FIND
END

```

プログラム **WTLEX1** の出力 - ページ 1 :

```

09:36:09.5  PEOPLE LIVING IN BARCELONA                PAGE:      1
-----
          NAME                FIRST-NAME                CURRENT
                               POSITION
-----
DEL CASTILLO          ANGEL                EJECUTIVO DE VENTAS
GARCIA                M. DE LAS MERCEDES  SECRETARIA
GARCIA                ENDIKA              DIRECTOR TECNICO
MARTIN                ASUNCION            SECRETARIA
MARTINEZ              TERESA              SECRETARIA
YNCLAN                FELIPE              ADMINISTRADOR
FERNANDEZ            ELOY                OFICINISTA
TORRES                ANTONI              OBRERA

CITY OF BARCELONA REGISTER
-----

```

プログラム **WTLEX1** の出力 - ページ 2 :

```

09:37:26.0  PEOPLE LIVING IN BARCELONA                PAGE:      2
-----
          NAME                FIRST-NAME                CURRENT
                               POSITION
-----
RODRIGUEZ            VICTORIA            SECRETARIA
GARCIA                GERARDO             INGENIERO DE PRODUCCION

CITY OF BARCELONA REGISTER
-----

```





# 133

## WRITE WORK FILE

---

■ 機能 .....	904
■ 構文説明 .....	904
■ フィールドの外部表示 .....	905
■ ラージおよびダイナミック変数の処理 .....	906
■ 例 .....	907

```
WRITE WORK [FILE] work-file-number [VARIABLE] operand1 ...
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連ステートメント：[DEFINE WORK FILE](#) | [READ WORK FILE](#) | [CLOSE WORK FILE](#) | [DOWNLOAD PC FILE](#)

関連機能グループ：「[ワークファイル/PC ファイルの制御](#)」

## 機能

WRITE WORK FILE ステートメントは、物理順ワークファイルにレコードを書き込むために使用します。

このステートメントは、Com-plete、CICS、CMS、TSO、TIAM、またはバッチモード環境で実行されるプログラムでのみ使用できます。適切な JCL またはシステムコマンドを実行してワークファイルを割り当てる必要があります。詳細については、『[オペレーション](#)』ドキュメントを参照してください。ワークファイル割り当ての詳細については、『[パラメータリファレンス](#)』ドキュメントでプロファイルパラメータ WORK を参照してください。

1つのプログラム内または処理ループ内でワークファイルを作成し、そのファイルを後続の別の処理ループまたはプログラムで [READ WORK FILE](#) ステートメントを使用して読み取ることもできます。



**注意:** Unicode およびコードページのサポートについては、『[Unicode およびコードページのサポート](#)』ドキュメントの「[メインフレームプラットフォーム上のワークファイルと出力ファイル](#)」を参照してください。

## 構文説明

オペランド定義テーブル：

オペランド	構文要素	フォーマット	ステートメント参照	ダイナミック定義
<i>operand1</i>	C S A G	A U N P I F B D T L C G	可	不可



**注意:** Natural Connection ではフォーマット C もフォーマット G も使用できません。

構文要素の説明：

<i>work-file-number</i>	<p>ワークファイル番号：</p> <p>使用するワークファイル番号（Natural に定義されたもの）を指定します。</p>
<b>VARIABLE</b>	<p>可変エントリ：</p> <p>同一ワークファイルに対し、異なる WRITE WORK FILE ステートメントで、異なるフィールドを持つレコードを書き込むことができます。この場合、すべてのWRITE WORK FILE ステートメントに VARIABLE を指定する必要があります。外部ファイルに書き込まれるレコードは可変フォーマットです。Natural は出力ファイルをすべて可変ブロックとして書き出します（ユーザーが実行JCLにレコードフォーマットとブロックサイズを指定しない場合）。</p> <p>オペランドリストにダイナミック変数（WRITE WORK FILE ステートメントの異なる実行に対してサイズが変わる）が含まれている場合、VARIABLE エントリはすべての WRITE WORK FILE ステートメントに指定する必要があります。</p> <p>変数インデックス範囲：</p> <p>ワークファイルに配列を書き込む場合、配列に変数インデックス範囲を指定できます。次に例を示します。</p> <pre>WRITE WORK FILE <i>work-file-number</i> VARIABLE #ARRAY (I:J)</pre>
<i>operand1</i>	<p>フィールド：</p> <p><i>operand1</i>では、ワークファイルに書き込むフィールドを指定します。ワークファイルに書き込むフィールドには、データベースフィールド、ユーザー定義変数、または READ WORK FILE ステートメントを使用して別のワークファイルから読み取ったフィールドを指定できます。</p> <p>データベースの配列は、ワークファイルに書き込むオカレンスを示すインデックスの単一範囲で参照できます。データベースファイルのグループは、グループ名で参照できます。グループに属するすべてのデータベースフィールドがワークファイルに書き込まれます。</p>


## フィールドの外部表示

WRITE WORK FILE ステートメントで書き込まれるフィールドは、その内部定義に従って外部ファイルに表示されます。フィールド値は編集されません。

A および B フォーマットのフィールドについては、外部ファイルでのバイト数は、Natural プログラムで定義された内部長さ定義と同じです。値は編集されず、小数点は表示されません。


N フォーマットのフィールドについては、外部ファイルでのバイト数は、小数点の前後の内部桁数の合計です。小数点は外部ファイルには表示されません。

Pフォーマットのフィールドについては、外部ファイルでのバイト数は、小数点の前後の桁数の合計に1（符号用）を加算し、それを2で割ってバイト単位に切り上げたものです。

 **注意:** フォーマット変換は、ワークファイルに書き込まれるフィールドに対しては行われません。

フィールド表現の例：

フィールド定義	出力レコード
#FIELD1 (A10)	10 バイト
#FIELD2 (B15)	15 バイト
#FIELD3 (N1.3)	4 バイト
#FIELD4 (N0.7)	7 バイト
#FIELD5 (P1.2)	2 バイト
#FIELD6 (P6.0)	4 バイト

 **注意:** 数値フィールド（フォーマットNまたはP）に対するNaturalシステム関数AVER、NAVER、SUM、またはTOTALをワークファイルに書き込むとき、これらのフィールドの内部長は1桁拡張されます（例：フォーマットP3のフィールドのSUMはP4に拡張されます）。これにより、ワークファイルを読み取るときに注意する必要があります。

## ラージおよびダイナミック変数の処理

ワークファイルタイプ	処理
<b>UNFORMATTED</b>	ワークファイルタイプ UNFORMATTED は、最大レコード長を超えるサイズの変数を書き込むために使用できます。「ラージ変数およびダイナミック変数によるワークファイルへのアクセス」も参照してください。
<b>FORMATTED</b>	ダイナミック変数は現在定義されている長さ（長さ0を含む）で書き込まれません。

## 例

```
** Example 'WFEX1': WRITE WORK FILE
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
END-DEFINE
*
FIND EMPLOY-VIEW WITH CITY = 'LONDON'
  WRITE WORK FILE 1
    PERSONNEL-ID NAME
END-FIND
*
END
```



# 134 SQL ステートメント

---

Natural DML ステートメントの他に、Natural プログラムで SQL ステートメントを使用して直接 SQL を扱うことができます。

次の SQL ステートメントが有効です。

**CALLDBPROC | COMMIT | DELETE | INSERT | PROCESS SQL | READ RESULT SET  
| ROLLBACK | SELECT | UPDATE**



**注意:** Natural アプリケーションの移植性に関して、Natural がサポートする全データベースシステムに対して使用できる FIND、READ のような Natural DML ステートメントとは別に、Natural SQL ステートメントは SQL 対応のデータベースシステムにだけ使用できることに注意してください。

このセクションでは、次のトピックについて説明します。

- 一般セットと拡張セット
- 基本構文項目
- **Natural** ビューの概念
- スカラー式
- 検索条件
- 選択式
- **フレキシブル SQL**
- SQL ステートメント - アルファベット順

「フレキシブル SQL」と呼ばれる SQL ステートメント発行の追加機能では、任意の SQL 構文を使用できます。





# 135 一般セットと拡張セット

---

Natural プログラミング言語内で使用できる SQL ステートメントは、次の2つの異なる構文セットから成ります。

## ■ 一般セット

基本的に標準の SQL 構文定義に対応しており、Natural がサポートする SQL 対応の各データベースシステム用に提供されています。

## ■ 拡張セット

サポートされている各種のデータベースシステム特有の機能をサポートするため、一般セットに特別な機能拡張を提供しています。拡張セットのサポート部分は各データベースシステムによって異なります。

Natural SQL ステートメントのドキュメントでは、主に Natural SQL 一般セットについて説明しています。ステートメント構文は、できる限り SQL の文献に記載されている構文に従っているため、これらの文献も参照してください。Natural SQL 拡張セットの詳細については、各データベースシステムに特有の Natural インターフェイスのドキュメントを参照してください。



# 136 基本構文項目

---

▪ 定数 .....	914
▪ 名前 .....	914
▪ パラメータ .....	917
▪ Natural フォーマットと SQL データタイプ .....	920

この章では、個々の SQL ステートメントの説明には詳しく記載されていない基本構文項目について説明します。

## 定数


---

Natural SQL ステートメントの構文記述で使用される定数は次のとおりです。

- *constant*
- *integer*

これらの各項目については、以下で説明します。

<i>constant</i>	項目 <i>constant</i> は、常に Natural 定数を表します。
<i>integer</i>	項目 <i>integer</i> は、常に整数の定数を表します。

 **注意:** 小数点表記の文字（セッションパラメータ DC）がコンマ (,) に設定されている場合、指定した数値定数の直後にコンマを続けてはならず、定数とコンマを空白文字で区切る必要があります。そうしないと、エラーまたは不正な結果が生じます。

無効な構文：	有効な構文：
VALUES (1,'A') leads to a syntax error	VALUES (1 , 'A')
VALUES (1,2,3) leads to wrong results	VALUES (1 ,2 ,3)

## 名前

---

Natural SQL ステートメントの構文記述で使用される名前は次のとおりです。

- *authorization-identifier*
- *dsm-name*
- *view-name*
- *column-name*
- *table-name*
- *correlation-name*

これらの各項目については、以下で説明します。

<i>authorization-identifier</i>	項目 <i>authorization-identifier</i> (作成者名とも呼ばれる) は、データベーステーブルとビューを修飾するために使用します。下記も参照してください。
<i>ddm-name</i>	項目 <i>ddm-name</i> は、常に Natural ユーティリティ SYSDDM で作成された Natural DDM の名前を表します。
<i>view-name</i>	項目 <i>view-name</i> は、常に DEFINE DATA ステートメントで定義された Natural ビューの名前を表します。
<i>column-name</i>	項目 <i>column-name</i> は、常に物理データベース列の名前を表します。
<i>table-name</i>	<p>構文：</p> <pre><i>authorization-identifier</i><i>ddm-name</i></pre> <p>このセクションの項目 <i>table-name</i> は、SQL 基本テーブルと SQL ビューテーブルの両方を参照するために使用します。Natural DDM は、使用するテーブル用に作成されている必要があります。このような DDM の名前は、対応するデータベーステーブル名またはビュー名と同じにする必要があります。</p> <p><i>authorization-identifier</i></p> <p>データベーステーブルまたはビューの <i>authorization-identifier</i> は、2 とおりの方法で指定できます。</p> <p>1 つの方法は標準 SQL 構文に対応しています。この場合、<i>authorization-identifier</i> とテーブル名はピリオドで区切ります。この形式を使用する場合、<i>authorization-identifier</i> を使用せずに、DDM の名前をデータベーステーブルの名前と同じにする必要があります。</p> <p>例：</p> <pre>DEFINE DATA LOCAL 01 PERS VIEW OF PERSONNEL    02 NAME    02 AGE END-DEFINE SELECT *   INTO VIEW PERS   FROM SQL.PERSONNEL ...</pre> <p>別の方法として、<i>authorization-identifier</i> を DDM 名の一部として定義できます。DDM 名は、<i>authorization-identifier</i> とデータベーステーブル名をハイフン (-) で区切って構成されます。<i>authorization-identifier</i> とテーブル名の間ハイフンは内部的にピリオドに変換されます。</p>

	<p><b>注意:</b> この形式の DDM 名は、<b>FIND</b> または <b>READ</b> ステートメントでも使用できます。これらのステートメントに適用される DDM 命名規則に準拠しているためです。</p> <p>例:</p> <pre> DEFINE DATA LOCAL 01 PERS VIEW OF SQL-PERSONNEL     02 NAME     02 AGE END-DEFINE SELECT *     INTO VIEW PERS     FROM SQL-PERSONNEL ... </pre> <p><i>authorization-identifier</i> が明示的にも DDM 名内にも指定されていない場合、SQL データベースシステムによって決定されます。</p> <p><i>table-names</i> は、<b>SELECT</b> ステートメントで使用されるほか、<b>DELETE</b>、<b>INSERT</b>、および <b>UPDATE</b> ステートメントにも指定できます。</p> <p>例:</p> <pre> ... DELETE FROM SQL.PERSONNEL     WHERE AGE IS NULL ...  ... INSERT INTO SQL.PERSONNEL (NAME,AGE)     VALUES ('ADKINSON',35) ...  ... UPDATE SQL.PERSONNEL SET SALARY = SALARY * 1.1 WHERE AGE &gt; 30 ... </pre>
<p><i>correlation-name</i></p>	<p>項目 <i>correlation-name</i> は、テーブル名のエイリアス名を表します。列名を修飾するために使用できます。また、<b>SELECT</b> ステートメントの <b>INTO</b> 節で使用すると、Natural ビューでフィールドを暗黙的に修飾できます。</p>

	<p>例：</p> <pre> DEFINE DATA LOCAL 01 PERS-NAME      (A20) 01 EMPL-NAME      (A20) 01 AGE            (I2) END-DEFINE ... SELECT X.NAME , Y.NAME , X.AGE       INTO PERS-NAME , EMPL-NAME , AGE       FROM SQL-PERSONNEL X , SQL-EMPLOYEES Y       WHERE X.AGE = Y.AGE END-SELECT ... </pre> <p>ほとんどの場合、<i>correlation-names</i>を使用する必要はありませんが、使用するとステートメントがわかりやすくなります。</p>
--	---

## パラメータ

*parameter*

`[:] host-variable [INDICATOR [:] host-variable] [LINDICATOR [:] host-variable]`

構文の各項目については、以下で説明します。

<i>host-variable</i>	<p><i>host-variable</i>は、SQLステートメントで参照される Natural ユーザー定義変数（システム変数ではない）です。個々のフィールドまたは Natural ビューの一部として定義できます。</p> <p>受信側フィールドとして（INTO 節などで）定義する場合、<i>host-variable</i>では値がデータベースシステムによって割り当てられている変数を識別します。</p> <p>送信側フィールドとして（WHERE 節などで）定義する場合、<i>host-variable</i>ではプログラムからデータベースシステムに渡される値を指定します。</p> <p>「<a href="#">Natural フォーマットと SQL データタイプ</a>」も参照してください。</p>
[:]	<p>コロン：</p> <p>SQL 標準に準拠するために、ホスト変数の先頭にコロン（:）を付けることもできます。フレキシブル SQL で使用する場合、ホスト変数はコロンで修飾する必要があります。</p>

	<p>例：</p> <pre>SELECT NAME INTO :#NAME FROM PERSONNEL WHERE AGE = :VALUE</pre> <p>変数名が SQL 予約語と同一の場合、コロンは常に必要です。ホスト変数または列の参照が可能な状況では、コロンなしの名前を使用すると、列への参照として解釈されます。</p>
<p><b>INDICATOR</b></p>	<p><b>INDICATOR 節：</b></p> <p>INDICATOR 節は、「空値」（つまり、値なし）と実際の値 0 または「空白」を区別するためのオプション機能です。</p> <p>受信側 <i>host-variable</i>（ターゲットフィールド）で指定する場合、INDICATOR <i>host-variable</i>（空値インジケータフィールド）によって、取得される列が「空値」であるかどうかを調べることができます。</p> <p>例：</p> <pre>DEFINE DATA LOCAL 1 NAME      (A20) 1 NAMEIND   (I2) END-DEFINE SELECT *   INTO NAME INDICATOR NAMEIND ...</pre> <p>この例では、NAME は受信側 <i>host-variable</i> を表し、NAMEIND は空値インジケータフィールドを表します。</p> <p>空値インジケータフィールドが指定されており、取得される列が空値の場合、空値インジケータフィールドの値は負になり、そのデータタイプに応じてターゲットフィールドが 0 または「空値」に設定されます。それ以外の場合、空値インジケータフィールドの値は 0 以上になります。</p> <p>送信側 <i>host-variable</i>（ソースフィールド）で指定する場合、空値インジケータフィールドはこのフィールドに空値を指定するために使用されます。</p> <p>例：</p> <pre>DEFINE DATA LOCAL 1 NAME      (A20) 1 NAMEIND   (I2) UPDATE ... SET NAME = :NAME INDICATOR :NAMEIND WHERE ...</pre> <p>この例では、:NAME は送信側ホスト変数を表し、:NAMEIND は空値インジケータフィールドを表します。空値インジケータフィールドの入力として負の値を入力すると、空値がデータベース列に割り当てられます。</p>



	INDICATOR <i>host-variable</i> のフォーマット／長さは I2 です。
LINDICATOR	<p><b>LINDICATOR 節：</b></p> <p>LINDICATOR 節は、可変長（例えば、VARCHAR または LONG VARCHAR タイプ）の列をサポートするためのオプション機能です。</p> <p>受信側 <i>host-variable</i>（ターゲットフィールド）で指定する場合、LINDICATOR <i>host-variable</i>（長さインジケータフィールド）にはデータベースによってターゲットフィールドに実際に返された文字数が入ります。ターゲットフィールドは常に空白で埋められます。</p> <p>VARCHAR または LONG VARCHAR 列にターゲットフィールドに収まりきらない文字が含まれている場合、長さインジケータフィールドは実際に返された長さ（つまり、ターゲットフィールドの長さ）に設定され、空値インジケータフィールド（指定されている場合）はこの列の合計長に設定されます。</p> <p>例</p> <pre>DEFINE DATA LOCAL 1 ADDRESSLIND (I2) 1 ADDRESS      (A50/1:6) END-DEFINE SELECT *   INTO :ADDRESS(*) LINDICATOR :ADDRESSLIND   ...</pre> <p>この例では、:ADDRESS(*) はアドレス指定された VARCHAR または LONG VARCHAR 列の最初の 300 バイト（利用可能な場合）を受け取るターゲットフィールドを表し、:ADDRESSLIND は実際に返された文字数を含む長さインジケータフィールドを表します。</p> <p>送信側 <i>host-variable</i>（ソースフィールド）で指定する場合、長さインジケータフィールドではデータベースに渡されるソースフィールドの文字数を指定します。</p> <p>例：</p> <pre>DEFINE DATA LOCAL 1 NAMELIND (I2) 1 NAME      (A20) 1 AGE       (I2) END-DEFINE MOVE 4      TO NAMELIND MOVE 'ABC%' TO NAME SELECT AGE   INTO :AGE WHERE NAME LIKE :NAME LINDICATOR :NAMELIND   ...</pre> <p>LINDICATOR <i>host-variable</i> のフォーマット／長さは I2 または I4 です。パフォーマンス上の理由から、対応するターゲットフィールドまたはソースフィールドの直前に指</p>

	<p>定する必要があります。そうしないと、フィールドはランタイムに一時ストレージにコピーされます。</p> <p>LINDICATOR フィールドを I2 フィールドとして定義すると、対応する列の送受信には SQL データタイプ VARCHAR が使用されます。LINDICATOR <i>host-variable</i> を I4 として指定すると、ラージオブジェクトデータタイプ (CLOB/BLOB) が使用されます。</p> <p>フィールドを DYNAMIC として定義すると、列は内部ループでは実際の長さまで読み取られます。LINDICATOR フィールドと *LENGTH はこの長さに設定されます。固定長フィールドの場合、列は定義された長さまで読み込まれます。いずれの場合も、フィールドは LINDICATOR フィールドで定義した値まで書き込まれます。</p> <p>固定長フィールドの定義には、I2 として指定した LINDICATOR フィールドを使用してください。VARCHAR 列にこの固定長フィールドに収まりきらない文字が含まれている場合、長さインジケータフィールドは実際に返された長さに設定され、空値インジケータフィールド (指定されている場合) はこの列の合計長に設定されます (取得)。これは 32KB 以上の固定長フィールドには適用されません (長さが空値インジケータフィールドに収まらないため)。</p>
--	--

## Natural フォーマットと SQL データタイプ

ホスト変数の Natural フォーマットは、次の表に従って SQL データタイプに変換されます。

Natural フォーマット / 長さ	SQL データタイプ
A $n$	CHAR ( $n$ )
B2	SMALLINT
B4	INT
B $n$ ( $n$ は 2 または 4 以外)	CHAR ( $n$ )
F4	REAL
F8	DOUBLE PRECISION
I2	SMALLINT
I4	INT
N $nn.m$	NUMERIC ( $nn+m, m$ )
P $nn.m$	NUMERIC ( $nn+m, m$ )
T	TIME
D	DATE
G $n$ (ビューフィールドのみ)	GRAPHIC ( $n$ )

Natural では、変換後の SQL データタイプがデータベース列と互換性があるかどうかはチェックされません。フォーマット N のフィールドを除き、データ変換は行われません。

また、Natural SQL では、標準 Natural フォーマットに対する次の拡張を使用できます。

- フォーマット A の 1 次元配列を使用して、253 バイトを超える英数字列をサポートできます。この配列はインデックス 1 から定義する必要があり、インデックスとしてアスタリスク (\*) を使用することでのみ参照できます。対応する SQL データタイプは CHAR (n) です。ここで、n は配列の合計バイト数です。
- キーワード LINDICATOR で指定した特別なホスト変数を使用して、可変長列をサポートできます。対応する SQL データタイプは VARCHAR (n) です。LINDICATOR 節も参照してください。
- Natural for DB2 では、Natural フォーマットの日付 (D) および時刻 (T) を使用できます。これらのフォーマットは、DB2 DATE および TIME に変換されます (『データベース管理システムインターフェイス』ドキュメントの *Natural for DB2* に関する説明も参照)。

LINDICATOR フィールドを使用せずに 1 次元配列として指定した送信側フィールドは、SQL データタイプ VARCHAR に変換されます。長さは配列の合計バイト数になり、末尾の空白は考慮されません。



# 137 Natural ビューの概念

いくつかの Natural SQL ステートメントも Natural ビューの使用をサポートします。

Natural ビューをパラメータリストの代わりに指定できます。ビューの各フィールド（グループフィールド、再定義フィールド、および L@ や N@ の接頭文字付きフィールド以外）は 1 パラメータ（ホスト変数）に対応します。

名前に接頭文字 L@ や N@ が付加されたフィールドは、対応するマスタフィールド（同じ名前のフィールド）とともにのみ存在させることができます。

- L@ フィールドは LINDICATOR フィールドに変換されます。
- N@ フィールドは INDICATOR フィールドに変換されます。

ビュー定義で、L@ フィールドは、適用先のマスタフィールドの直前に指定してください。

```
DEFINE DATA LOCAL
01 PERS VIEW OF SQL-PERSONNEL
  02 PERSID      (I4)
  02 NAME       (A20)
  02 N@NAME     (I2)                /* null indicator of NAME
  02 L@ADDRESS  (I2)                /* length indicator of ADDRESS
  02 ADDRESS    (A50/1:6)
  02 N@ADDRESS  (I2)                /* null indicator of ADDRESS
01 #PERSID      (I4)
END-DEFINE
...
SELECT *
  INTO VIEW PERS
  FROM SQL-PERSONNEL
  WHERE PERSID = #PERSID
...
END-SELECT
```


上記の例は下記の例と同等です。

```
...  
SELECT *  
  INTO PERSID,  
      NAME INDICATOR N@NAME,  
      ADDRESS(*)INDICATOR N@ADDRESS LINDICATOR L@ADDRESS  
  FROM SQL-PERSONNEL  
  WHERE PERSID = #PERSID  
...  
END-SELECT
```

# 138 スカラー式

---

■ スカラー式 .....	926
■ スカラー演算子 .....	926
■ ファクタ .....	927



## スカラー式

---


*scalar-expression*は、ファクタまたはスカラー演算子を含む他のスカラー式で構成されます。

参照の優先順位に関して、スカラー式は次のように処理されます。

- 非修飾変数名がスカラー式に指定されるとき、最初に変数名は参照テーブルの列名として解決されます。
- 指定された名前が参照テーブルで有効でない場合、Naturalはこの変数をNaturalユーザー定義変数（ホスト変数）として解決しようとします。

## スカラー演算子

---



*scalar-operator*は上記のいずれかの演算子です。マイナス (-) およびスラッシュ (/) の演算子を使用する場合は、先行する演算子との間を最低1つの空白で区切る必要があります。



## ファクタ

一般セットの構文：

```
{  
  atom  
  column-reference  
  aggregate-function  
  special-register  
}
```

拡張セットの構文：

```
atom  
{  
  column-reference  
  aggregate-function  
  special-register  
  スカラー関数 (スカラー式,...)  
  スカラー式の単位  
  CASE 式  
  CAST 式  
  ユーザー定義関数参照  
  シーケンス参照  
}
```

*factor* は、上図のいずれかの項目で構成されます。これらの項目について次に説明します。

### 原子


```
{  
  parameter  
  constant  
}
```

*atom* は、*parameter* または *constant* です。「[基本構文項目](#)」も参照してください。

## 列参照

```
[ table-name .
  correlation-name ] column-name
```

*column-reference* は、オプションで *table-name* または *correlation-name* (「[基本構文項目](#)」も参照) で修飾された列名です。修飾された名前は修飾されない名前よりも明確であり、ときには修飾が必要となります。

 **注意:** このような場合には、テーブル名を認可 ID で明示的に修飾しないでください。テーブル名を修飾するときは、代わりに相関名を使用してください。

*table-name* または *correlation-name* で列を参照する場合、列は対応するテーブル内に含まれている必要があります。 *table-name* も *correlation-name* も指定されていない場合、各列は FROM 節に指定されたテーブルの 1 つに存在している必要があります。

## 集積関数

一般セットの構文:

```
{ COUNT { (*)
      (DISTINCT column-reference) }
  { AVG
    { MAX { (DISTINCT column-reference) }
      MIN { ([ALL] scalar-expression) }
      SUM }
```

拡張セットの構文:

COUNT	{ (*) (DISTINCT <i>column-reference</i> ) }
AVG	{ (DISTINCT <i>column-reference</i> ) ([ALL] <i>scalar-expression</i> ) }
MAX	
MIN	
SUM	
COUNT_BIG	
STDDEV	
STDDEV_POP	
STDDEV_SAMP	
VAR	
VAR_POP	
VAR_SAMP	
VARIANCE	
VARIANCE_SAMP	

SQL は基本的な取得機能拡張のために多くの特別関数を提供しています。現在有効で Natural がサポートする、いわゆる SQL 集積関数は次のとおりです。

<b>AVG</b>	列の平均値
<b>COUNT</b>	列の値の個数
<b>MAX</b>	列の最大値
<b>MIN</b>	列の最小値
<b>SUM</b>	列の合計値

COUNT(\*) は別にして、上記の関数は引数（つまり、単一の列または *scalar-expression*）のスカラー値の集合を処理します。その結果としてスカラー値を生成します。

例：

```
DEFINE DATA LOCAL
1  AVGAGE   (I2)
END-DEFINE
...
SELECT AVG (AGE)
  INTO  AVGAGE
  FROM  SQL-PERSONNEL
...
```

関数が適用される前に余分な重複値を削除するため、一般的に、引数の前にキーワード DISTINCT をオプションで指定できます。

## スカラー式

---

DISTINCTを指定する場合、引数は単一の列名にする必要があります。DISTINCTを指定しない場合、引数は一般的な *scalar-expression* の構成にすることもできます。

重複値を削除せずにテーブルのすべての行をカウントする特別関数 COUNT(\*) に DISTINCT を指定することはできません。

### 特別レジスタ

一般セットの構文：

```
USER
```

拡張セットの構文：

```
USER  
{  
CURRENT TIMEZONE  
CURRENT DATE  
CURRENT TIME  
CURRENT TIMESTAMP  
CURRENT SQLID  
CURRENT PACKAGESET  
CURRENT SERVER  
}
```

*special-register* に対する参照はスカラー値を返します。

USERを除き、*special-registers* は標準 SQL には準拠しておらず、Natural SQL 拡張セットでのみサポートされています。

### スカラー関数

*scalar-function* は、スカラー計算式の構築に使用できる組み込み関数です。

Natural SQL 拡張セットでサポートされているスカラー関数の詳細については、『Natural for DB2』ドキュメントの「Natural SQL Statements - Syntactical Items」にある「*scalar-function*」を参照してください。

## スカラー式の単位

YEAR
YEARS
MONTH
MONTHS
DAY
DAYS
HOUR
HOURS
MINUTE
MINUTES
SECOND
SECONDS
MICROSECOND
MICROSECONDS

*unit* は標準 SQL には準拠しておらず、Natural SQL 拡張セットでのみサポートされています。

## CASE 式

```
CASE { searched-when-clause ... } [ ELSE { NULL scalar-expression } ] END
```

*case-expression* は標準 SQL には準拠しておらず、Natural SQL 拡張セットでのみサポートされています。

## 検索 WHEN 節

```
WHEN search-condition THEN { NULL scalar-expression }
```

検索 WHEN 節は標準 SQL には準拠しておらず、Natural SQL 拡張セットでのみサポートされています。

詳細については、*search-condition* を参照してください。

### 単純 WHEN 節

```
scalar-expression { WHEN scalar-expression THEN { NULL  
scalar-expression } }...
```

単純 WHEN 節は標準 SQL には準拠しておらず、Natural SQL 拡張セットでのみサポートされています。

### CAST 式

```
CAST (scalar-expression AS data-type)
```

CAST 式は標準 SQL には準拠しておらず、Natural SQL 拡張セットでのみサポートされています。

### ユーザー定義関数参照

オプション *user-defined-function-reference* は Natural SQL 拡張セットに含まれます。このオプションでは、ユーザー定義関数を呼び出すことができます。引数は角カッコ内にコンマで区切って指定する必要があります。ユーザー定義関数はターゲット RDBMS で宣言する必要があります。

### シーケンス参照

```
NEXT VALUE FOR sequence-name  
PREVIOUS VALUE FOR sequence-name
```

オプション *sequence-reference* は Natural SQL 拡張セットに含まれます。

このオプションでは、シーケンスオブジェクトの次の値または前の値を参照できます。シーケンスオブジェクトは、ランタイムに参照される前に、ターゲット RDBMS で作成する必要があります。

### スカラー全選択

```
(fullselect)
```

オプション *scalar-fullselect* は Natural SQL 拡張セットに含まれます。


式では、単一の列値で構成される単一行を返す *scalar-fullselect* (カッコで囲んだ *fullselect*) がサポートされています。 *fullselect* で行が返されない場合、式の結果は空値になります。 *scalar-fullselect* に対して複数の行が返される場合、エラーが発生します。

# 139 検索条件

---

▪ 検索条件 .....	934
▪ 属性 .....	934

## 検索条件




## 検索条件

*search-condition* は、単純な *predicate*、またはブール演算子 AND、OR、NOT、およびカッコ（評価の順序を示す必要がある場合）を組み合わせた複数の *search-conditions* で構成できます。

例

```
DEFINE DATA LOCAL
01 NAME (A20)
01 AGE (I2)
END-DEFINE
...
SELECT *
  INTO NAME, AGE
  FROM SQL-PERSONNEL
  WHERE AGE = 32 AND NAME > 'K'
END-SELECT
...
```

## 属性





**EXISTS subquery**

*predicate*には「true」（真）、「false」（偽）、「unknown」（不明）の条件を指定します。

*search-condition*の*predicate*は、単純な比較演算または複雑な比較演算、あるいは他の種類の条件で構成できます。

例：

```
SELECT NAME, AGE
  INTO VIEW PERS
  FROM SQL-PERSONNEL
  WHERE AGE BETWEEN 20 AND 30
        OR AGE IN ( 32, 34, 36 )
        AND NAME LIKE '%er'
  ...
```



**注意:** パーセント記号（%）はNatural 端末コマンドと混同するおそれがあります。その場合は、%と異なる端末コマンド制御文字を定義する必要があります。

各属性については、以降のトピックで説明します（属性の詳細については、関連文献を参照してください）。上記の構文に関する属性は次のように呼ばれます。

- 比較属性
- BETWEEN 属性
- LIKE 属性
- NULL 属性
- IN 属性
- 数量属性
- EXISTS 属性


## 比較属性

$  \text{scalar-expression comparison} \left\{ \begin{array}{l} \text{scalar-expression} \\ \text{subquery} \end{array} \right\}  $
---

比較属性は2つの値を比較します。

詳細については、*scalar-expression*を参照してください。

## 比較



*comparison* は次のいずれかの演算子です。

=	等しい
<	より小さい
>	より大きい
<=	より小さいまたは等しい
>=	より大きいまたは等しい
<>	等しくない
!=	等しくない
>	以下
<	以上

## サブクエリ

`(select-expression)`

*subquery* は同様の他の式内でネストされた *select-expression* です。

例：

```
DEFINE DATA LOCAL
1 #NAME      (A20)
1 #PERSNR    (I4)
END-DEFINE
...
SELECT NAME, PERSNR
  INTO #NAME, #PERSNR
  FROM SQL-PERSONNEL
  WHERE PERSNR IN
```

```
( SELECT PERSNR
   FROM SQL-AUTOMOBILES
   WHERE COLOR = 'black' )
...
END-SELECT
```

詳細については、「[選択式](#)」を参照してください。

## BETWEEN 属性

```
scalar-expression [NOT] BETWEEN scalar-expression AND scalar-expression
```

BETWEEN 属性は範囲値で値を比較します。

詳細については、[scalar-expression](#) を参照してください。

## LIKE 属性

```
column-reference [NOT] LIKE { atom
                               special-register } [ESCAPE atom]
```

LIKE 属性は特定のパターンを持つ文字列を検索します。

詳細については、[column-reference](#)、[atom](#)、および[special-register](#) を参照してください。

## NULL 属性

```
column-reference IS [NOT] NULL
```

NULL 属性は空値のテストを行います。

詳細については、[column-reference](#) を参照してください。

## IN 属性

```
scalar-expression { subquery
[NOT] IN             ( { atom
                       special-register }, ... ) }
```

IN 属性は値の集合で値を比較します。

詳細については、[scalar-expression](#)、[atom](#)、および[special-register](#) を参照してください。

詳細については、[subquery](#) を参照してください。

### 数量属性

```
scalar-expression comparison { ALL ANY SOME } subquery
```

数量属性は値の集合で値を比較します。

詳細については、*scalar-expression*、*comparison*、および *subquery* を参照してください。

### EXISTS 属性

```
EXISTS subquery
```

EXISTS 属性は特定の行の存在についてテストを行います。

EXISTS 属性は、*subquery* の評価結果が空でない場合、つまり、*subquery* の WHERE 節の検索条件を満たす *subquery* の FROM テーブルに最低 1 つのレコード（行）が存在する場合にのみ、真であると評価します。

EXISTS の例：

```
DEFINE DATA LOCAL
1 #NAME      (A20)
END-DEFINE
...
SELECT NAME
  INTO #NAME
  FROM SQL-PERSONNEL
  WHERE EXISTS
    ( SELECT *
      FROM SQL-EMPLOYEES
      WHERE PERSNR > 1000
        AND NAME < 'L' )
    ...
END-SELECT
...
```

詳細については、*subquery* を参照してください。

# 140 選択式

---

■ 選択 .....	940
■ テーブル式 .....	941

## 選択式

---

```
SELECT selection table-expression
```

*select-expression* では、結果テーブルを指定します。次のステートメントで使用されます。

**INSERT | SELECT**

## 選択

---

```
[ ALL | DISTINCT ] { { scalar-expression [[AS] correlation-name], ... } * }
```

*selection* では、選択する項目を指定します。

### ALL/DISTINCT

*select-expression* の結果から重複行が自動的に取り除かれることはありません。取り除く処理を要求するには、キーワード **DISTINCT** を指定します。

**DISTINCT** に代わるのが **ALL** です。指定がない場合は **ALL** とみなされます。

### スカラー式

単純な列名の代わりに、または列名とともに、スカラー演算子および計算値を与えるスカラー関数を含む一般的な *scalar-expressions* を含めることもできます（「**スカラー式**」も参照）。

例：

```
SELECT NAME, 65 - AGE
FROM SQL-PERSONNEL
...
```

### 相関名

結果の列に対するエイリアス名として、*scalar-expression* に *correlation-name* を割り当てることが可能です。

*correlation-name* は一意にする必要はありません。結果の列に *correlation-name* を指定しない場合、対応する *column-name* が使用されます（結果の列が列名から派生する場合。派生しない場合は結果テーブルは名前を持ちません）。結果の列の名前は、SELECT ステートメントの **ORDER BY** 節の列名などに使用できます。

## アスタリスク表記 - \*

FROM 節に指定された全テーブルのすべての列が選択されます。

例：

```
SELECT *
FROM SQL-PERSONNEL, SQL-AUTOMOBILES
...
```

## テーブル式

```
FROM table-reference,...
[WHERE search-condition]
[GROUP BY column-reference,...]
[HAVING search-condition]
```

*table-expression* では、どこから、どのような条件に従って行を選択するかを指定します。

### FROM 節

```
FROM table-reference,...
```

### テーブル参照

```
{ table-name [correlation-clause]
  [TABLE] subquery correlation-clause
  joined-table
  TABLE function-name (scalar-expression,...) correlation-clause }
```

FROM 節に指定したテーブルには、選択リストに使用した列が含まれている必要があります。

単一テーブルを指定するか、サブクエリまたは「ジョイン」処理（下記参照）による中間テーブルを生成できます。

さまざまなテーブル（つまり、DDM）を1つの FROM 節にアドレス指定できるだけでなく、*subqueries* を指定した場合は *table-expression* に複数の FROM 節を含めることができるため、関連する基準データベースの識別には、式全体の最初の FROM 節に指定した最初の DDM のデータベース ID (DBID) が使用されます。

## 選択式

TABLE *function-name* 節は SQL 拡張セットに属し、*column-name* リストを含む *correlation-clause* が必要になります。

*table-name* にはオプションで *correlation-clause* を割り当てることができます。 *subquery* の場合は、*correlation-clause* を割り当てする必要があります。

## 相関節

```
[AS] correlation-name [(column-name,...)]
```

*correlation-clause* は、キーワード AS (オプション) と *correlation-name* で構成され、オプションで *column-name* リストが続きます。 *column-name* リストは SQL 拡張セットに属します。

## ジョインテーブル

```
{ table-reference { INNER
                     { LEFT } OUTER
                     { RIGHT } JOIN table-reference ON join-condition
                     { FULL } } }
```

*joined-table* では、「ジョイン」処理による中間テーブルを指定します。

「ジョイン」には、INNER、LEFT OUTER、RIGHT OUTER、または FULL OUTER JOIN を使用できます。何も指定しないと、INNER が有効になります。

複数の「ジョイン」処理をネストできます。つまり、中間結果テーブルを作成するテーブル自体を JOIN (ジョイン) 処理または *subquery* の中間結果テーブルにすることができます。次に後者に *joined-table* を含めたり、その FROM 節に他の *subquery* を指定したりすることもできます。

## ジョイン条件

INNER、LEFT OUTER、および RIGHT OUTER ジョインの場合：

```
search-condition
```



FULL OUTER ジョインの場合：

```
full-join-expression = full-join-expression [AND ...]
```

## フルジョイン式

```
{ column-name
  { VALUE }
  { COALESCE } (column-name, ...) }
```

*join-expression* 内には、*column-names* および *scalar-function* VALUE（または同義語 COALESCE）のみを使用できます。詳細については、*column-name* を参照してください。

## WHERE 節

```
[WHERE search-condition]
```

WHERE 節を使用して行に対する選択条件 (*search-condition*) を指定します。

例：

```
DEFINE DATA LOCAL
01 NAME    (A20)
01 AGE     (I2)
END-DEFINE
...
SELECT *
  INTO NAME, AGE
  FROM SQL-PERSONNEL
  WHERE AGE = 32
END-SELECT
...
```

詳細については、*search-condition* を参照してください。

## GROUP BY 節

```
[GROUP BY column-reference, ...]
```

GROUP BY 節では、FROM 節に指定されたテーブルをグループに再配列します。これで、各グループ内のすべての行は GROUP BY 列に対して同じ値を持ちます。

選択リストの各 *column-reference* は、GROUP BY 列にするか、または *aggregate-function* 内に指定する必要があります。集積関数は個々のグループ（テーブル全体ではない）に適用されます。結果テーブルにはグループと同じ数の行が含まれます。

詳細については、[column-reference](#) および [aggregate-function](#) を参照してください。

例：

```
DEFINE DATA LOCAL
1 #AGE      (I2)
1 #NUMBER  (I2)
END-DEFINE
...
SELECT AGE , COUNT(*)
  INTO #AGE, #NUMBER
  FROM SQL-PERSONNEL
  GROUP BY AGE
  ...
```

GROUP BY 節の前に WHERE 節を指定した場合、グループ化が行われる前に WHERE 節を満たさないすべての行が削除されます。

### HAVING 節

[[HAVING search-condition](#)]

HAVING 節を指定する場合、GROUP BY 節も指定する必要があります。

WHERE 節を使用して結果テーブルから行を削除する場合のように、HAVING 節では *search-condition* を基準にしてグループを削除します。HAVING 節の *Scalar-expressions* はグループごとに単一の値にする必要があります。

詳細については、[scalar-expression](#) および [search-condition](#) を参照してください。

例：

```
DEFINE DATA LOCAL
1 #NAME      (A20)
1 #AVGAGE   (I2)
1 #NUMBER   (I2)
END-DEFINE
...
SELECT NAME, AVG(AGE), COUNT(*)
  INTO #NAME, #AVGAGE, #NUMBER
  FROM SQL-PERSONNEL
  GROUP BY NAME
  HAVING COUNT(*) > 1
  ...
```

# 141      フレキシブル SQL

---

- フレキシブル SQL の使用 ..... 946
- フレキシブル SQL でのテキスト変数の指定 ..... 947

## フレキシブル SQL の使用

SQL 構文（上記のセクションを参照）に加えて、フレキシブル SQL では任意の SQL 構文を使用できます。

文字 << および >>

フレキシブル SQL は << および >> 文字で囲まれます。任意の SQL テキストおよびホスト変数を指定できます。フレキシブル SQL 内では、ホスト変数の先頭にコロン (:) を付ける必要があります。

フレキシブル SQL 文字列は複数のステートメント行にわたって指定することもできます。コメントも指定できます（[PROCESS SQL](#) ステートメントも参照）。

フレキシブル SQL は、次の SQL 構文項目と置き換えて使用できます。

- *atom*
- *column-reference*
- *scalar-expression*
- *predicate*

フレキシブル SQL は、選択式の節間でも使用できます。

```
SELECT selection
<< ... >>
INTO ...
FROM ...
<< ... >>
WHERE ...
<< ... >>
GROUP BY ...
<< ... >>
HAVING ...
<< ... >>
ORDER BY ...
<< ... >>
```



**注意:** フレキシブル SQL に使用される SQL テキストは Natural コンパイラには認識されません。SQL テキスト（置き換えられたホスト変数とともに）は単に SQL 文字列にコピーされ、データベースシステムに渡されます。フレキシブル SQL の構文エラーは、データベースで対応するステートメントが実行されるときにランタイムに検出されます。

## 例 1 :

```
SELECT NAME
FROM SQL-EMPLOYEES
WHERE << MONTH ( BIRTH ) >> = << MONTH ( CURRENT_DATE ) >>
```

## 例 2 :

```
SELECT NAME
FROM SQL-EMPLOYEES
WHERE << MONTH ( BIRTH ) = MONTH ( CURRENT_DATE ) >>
```

## 例 3 :

```
SELECT NAME
FROM SQL-EMPLOYEES
WHERE SALARY > 50000
<< INTERSECT
  SELECT NAME
  FROM SQL-EMPLOYEES
  WHERE DEPT = 'DEPT10'
>>
```

## フレキシブル SQL でのテキスト変数の指定

フレキシブル SQL 内では、「テキスト変数」を指定することもできます。

```
<<:T:host-variable [LINDICATOR:host-variable]>>
```

構文の各項目については、以下で説明します。

<b>:T:</b>	<p>テキスト変数は接頭文字 :T: を付加した <i>host-variable</i> です。これは英数字フォーマットにする必要があります。</p> <p>SQL ステートメント内のテキスト変数は、ランタイムにその内容で置き換えられます。つまり、テキスト変数に含まれるテキスト文字列が SQL 文字列に挿入されるということです。</p> <p>置き換えの後、挿入されたテキスト文字列から末尾の空白が削除されます。</p> <p>テキスト変数の内容が構文的に正しい SQL 文字列であることをユーザー自身が確実にする必要があります。特に、テキスト変数の内容に <i>host-variables</i> が含まれないようにする必要があります。</p> <p>テキスト変数を含むステートメントは常にダイナミック SQL モードで実行されます。</p>
<b>LINDICATOR</b>	LINDICATOR オプション :

	<p>テキスト変数の後にキーワード LINDICATOR と長さインジケータ変数（先頭にコロンを付加した <i>host-variable</i>）を指定できます。</p> <p>長さインジケータ変数のフォーマット／長さは I2 にする必要があります。</p> <p>LINDICATOR 変数を指定しない場合、テキスト変数の内容全体が SQL 文字列に挿入されま す。</p> <p>LINDICATOR 変数を指定した場合、テキスト変数の内容の先頭 <math>n</math> (<math>n</math> は LINDICATOR 変数の 値になる) 文字のみが SQL 文字列に挿入されます。LINDICATOR 変数内の数がテキスト変 数の内容の長さより大きい場合、テキスト変数の内容全体が挿入されます。LINDICATOR 変 数内の数が負の値または 0 の場合は、何も挿入されません。</p> <p>一般的な情報については、<i>host-variable</i> を参照してください。</p>
--	--

### テキスト変数の使用例

```

DEFINE DATA LOCAL
01 TEXTVAR (A200)
01 TABLES VIEW OF SYSIBM-SYSTABLES
    02 NAME
    02 CREATOR
END-DEFINE
*
MOVE 'WHERE NAME > 'SYS' AND CREATOR = 'SYSIBM'' TO TEXTVAR
*
SELECT * INTO VIEW TABLES
FROM SYSIBM-SYSTABLES
<< :T:TEXTVAR >>
DISPLAY TABLES
END-SELECT
*
END
    
```

生成される SQL ステートメント（LISTSQL システムコマンドで表示される）は次のようになります。

```

SELECT NAME, CREATOR FROM SYSIBM.SYSTABLES:T: FOR FETCH ONLY
    
```

実行される SQL ステートメントは次のようになります。

```
SELECT TABNAME, CREATOR FROM SYSIBM.SYSTABLES  
WHERE TABNAME > 'SYS' AND CREATOR = 'SYSIBM'
```





# 142 CALLDBPROC - SQL

---

▪ 機能 .....	952
▪ 制限事項 .....	953
▪ 構文説明 .....	953
▪ 例 .....	955

```
CALLDBPROC dbprocddm-name

[USING]      [ parameter [ AD= { M } ] ] ...
              [           [ AD= { O } ] ] ...
              [           [ AD= { A } ] ] ...

[RESULT SETS] result-set...

[GIVING sqlcode]

[ CALLMODE= { NONE } ]
[ CALLMODE= { NATURAL } ]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[データベースへのアクセスと更新](#)」

『データベース管理システムインターフェイス』ドキュメントの「*Natural for DB2*」の「*NDB-CALLDBPROC*」も参照してください。

## 機能

CALLDBPROC ステートメントは、Natural からアクセスできる SQL データベースシステムのストアードプロシージャを呼び出すために使用します。

ストアードプロシージャは、Natural サブプログラムまたは Natural 以外のプログラミング言語で記述されたプログラムのいずれかです。

CALLDBPROC は、呼び出し側オブジェクトとストアードプロシージャ間のパラメータの受け渡しに加えて、「結果セット」をサポートします。結果セットはストアードプロシージャから呼び出し側オブジェクトへ、パラメータを使用するよりも大量のデータを返すことを可能にします。

結果セットは、ストアードプロシージャによって作成され、呼び出し側オブジェクトで `READ RESULT SET` ステートメントを使用して読み込んだり処理したりすることのできる「一時的な結果テーブル」です。



**注意:** 一般的に、ストアードプロシージャの呼び出しは Natural サブプログラムの呼び出しと比較されることがあります。CALLDBPROC ステートメントを実行するとき、制御はストアードプロシージャに渡され、ストアードプロシージャの処理後、呼び出し側オブジェクトに戻り、CALLDBPROC ステートメントの次のステートメントに処理が続きます。

## 制限事項

このステートメントは、Natural for SQL で使用することはできません。

## 構文説明

<i>dbproc</i>	<p>呼び出すストアードプロシージャ：</p> <p><i>dbproc</i> として、呼び出すストアードプロシージャの名前を指定します。その名前は英数字変数または定数（アポストロフィで囲む）として指定できます。</p> <p>名前はターゲットデータベースシステムのストアードプロシージャの命名規則に従う必要があります。</p> <p>ストアードプロシージャが <b>Natural</b> サブプログラムである場合、実際のプロシージャ名は 8 文字以下である必要があります。</p>				
<i>ddm-name</i>	<p><b>Natural</b> データ定義モジュールの名前：</p> <p>DDMの名前は、ストアードプロシージャを実行するデータベースの「アドレス」を示すように指定する必要があります。詳細については、「<i>ddm-name</i>」を参照してください。</p>				
<p>[USING]</p> <p><i>parameter</i></p>	<p>渡されるパラメータ：</p> <p><i>parameter</i> として、呼び出し側オブジェクトからストアードプロシージャに渡されるパラメータを指定できます。<i>parameter</i> には次のものを使用できます。</p> <ul style="list-style-type: none"> <li>■ ホスト変数（オプションで INDICATOR および LINDICATOR 節を使用）</li> <li>■ 定数</li> <li>■ キーワード NULL.</li> </ul> <p>詳細については、<i>host-variable</i> を参照してください。</p>				
AD=	<p>属性定義：</p> <p><i>parameter</i> が <i>host-variable</i> の場合、次のようにマークできます。</p> <table border="1" data-bbox="402 1486 1482 1827"> <tr> <td data-bbox="402 1486 906 1654">AD=O</td> <td data-bbox="906 1486 1482 1654"> <p>変更不可。セッションパラメータ AD=O を参照してください。</p> <p>(DB2 for z/OS での対応するプロシージャ表記：IN)</p> </td> </tr> <tr> <td data-bbox="402 1654 906 1827">AD=M</td> <td data-bbox="906 1654 1482 1827"> <p>変更可。セッションパラメータ AD=M を参照してください。</p> <p>(DB2 for z/OS での対応するプロシージャ表記：INOUT)</p> </td> </tr> </table>	AD=O	<p>変更不可。セッションパラメータ AD=O を参照してください。</p> <p>(DB2 for z/OS での対応するプロシージャ表記：IN)</p>	AD=M	<p>変更可。セッションパラメータ AD=M を参照してください。</p> <p>(DB2 for z/OS での対応するプロシージャ表記：INOUT)</p>
AD=O	<p>変更不可。セッションパラメータ AD=O を参照してください。</p> <p>(DB2 for z/OS での対応するプロシージャ表記：IN)</p>				
AD=M	<p>変更可。セッションパラメータ AD=M を参照してください。</p> <p>(DB2 for z/OS での対応するプロシージャ表記：INOUT)</p>				

	AD=A	<p>入力のみ。セッションパラメータ AD=A を参照してください。</p> <p>(DB2 for z/OS での対応するプロシージャ表記：OUT)</p>
	<p><i>parameter</i> が定数の場合は、AD を明示的に指定することはできません。定数には常に AD=0 が適用されます。</p>	
<i>result-set</i>	<p>結果セットロケータ変数用のフィールド：</p> <p><i>result-set</i> として、結果セットロケータを返すフィールドを指定します。</p> <p>結果セットには、フォーマット/長さ I4 の変数を指定する必要があります。</p> <p>結果セット変数の値は、結果セットを識別し、すぐ次の <b>READ RESULT SET</b> ステートメントで参照できる番号です。</p> <p><i>result-set</i> の値の順序は、ストアードプロシージャから返される結果セットの順序に対応します。</p> <p>結果セットの内容は、次の <b>READ RESULT SET</b> ステートメントで処理できます。</p> <p>結果セットが返されない場合は、対応する結果セット変数に 0 が含まれます。</p> <p>複数の結果セットを指定できます。</p> <p>「結果セット」（『データベース管理システムインターフェイス』ドキュメントの「<i>Natural for DB2</i>」）も参照してください。</p>	
<b>GIVING</b> <i>sqlcode</i>	<p><b>GIVING <i>sqlcode</i></b> オプション：</p> <p>このオプションは、ストアードプロシージャを呼び出す SQL CALL ステートメントの SQL コードを取得するために使用できます。</p> <p>このオプションを指定し、ストアードプロシージャの SQL コードが 0 でない場合、<i>Natural</i> エラーメッセージは発行されません。この場合、SQL コード値に対して取られる動作は、呼び出し側の <i>Natural</i> オブジェクト内でコード化されている必要があります。</p> <p><i>sqlcode</i> フィールドには、フォーマット/長さ I4 の変数を指定する必要があります。</p> <p><b>GIVING <i>sqlcode</i></b> オプションを指定せず、ストアードプロシージャの SQL コードが 0 でない場合は、<i>Natural</i> エラーメッセージが発行されます。</p>	
<b>CALLMODE</b>	<p><b>CALLMODE</b> パラメータ</p> <p>ストアードプロシージャが、PARAMETER STYLE GENERAL または PARAMETER STYLE GENERAL WITH NULL で定義された <i>Natural</i> サブプログラムの場合、CALLMODE=NATURAL を指定する必要があります。それ以外の場合は、NONE を指定します。</p> <p><b>注意:</b> CALLMODE=NATURAL には、ストアードプロシージャに渡される内部パラメータ、またストアードプロシージャから渡される内部パラメータにも影響します。詳細については、「CALLMODE=NATURAL」（『データベース管理システムインターフェイス』ドキュメントの「NDB - CALLDBPROC」）を参照してください。</p>	

## 例

次の例は、ストアードプロシージャ DEMO\_PROC を呼び出して、PERSON テーブルの特定範囲内の名前をすべて取得する Natural プログラムを示しています。

3つのパラメータフィールドが DEMO\_PROC に渡されます。1番目と2番目のパラメータは、名前の範囲の開始値と終了値をストアードプロシージャに渡します。3番目のパラメータは、条件に一致する名前を受け取ります。

この例では、名前は、**READ RESULT SET** ステートメントで処理される結果セットに返されます。

```

DEFINE DATA LOCAL
1 PERSON VIEW OF DEMO-PERSON
  2 PERSON_ID
  2 LAST_NAME
1 #BEGIN      (A2) INIT <'AB'>
1 #END        (A2) INIT <'DE'>
1 #RESPONSE  (I4)
1 #RESULT     (I4)
1 #NAME       (A20)
END-DEFINE

...

CALLDBPROC 'DEMO_PROC' DEMO-PERSON #BEGIN (AD=0) #END (AD=0) #NAME (AD=A)
  RESULT SETS #RESULT
  GIVING #RESPONSE

READ RESULT SET #RESULT INTO #NAME FROM DEMO-PERSON
  GIVING #RESPONSE
  DISPLAY #NAME
END-RESULT

...

END

```

他の例については、「*Example of CALLDBPROC/READ RESULT SET*」（『データベース管理システムインターフェイス』ドキュメントの「NDB - CALLDBPROC」）を参照してください。



# 143 COMMIT - SQL

---

▪ 機能 .....	958
▪ Natural 以外のプログラムについて .....	958
▪ 例 .....	958

**COMMIT**

関連機能グループ：「[データベースへのアクセスと更新](#)」


『データベース管理システムインターフェイス』ドキュメントの次のセクションも参照してください。

- 「*Natural for DB2*」の「*NDB - COMMIT*」
- 「*Natural for SQL/DS*」の「*COMMIT*」

---

## 機能

SQL COMMIT ステートメントは **END TRANSACTION** ステートメントに対応します。論理トランザクションの終了を示し、トランザクション中にロックされた全データを解放します。すべてのデータ変更がコミットされて確定されます。

 **重要:** 作業の論理ユニット終了時には全カーソルがクローズされるため、COMMIT ステートメントはデータベース更新処理ループ内に指定しないでください。更新処理ループの外側、またはネストされたループの一番外側のループの後に指定する必要があります。

---

## Natural 以外のプログラムについて

Natural プログラムから他の標準プログラミング言語で記述された外部プログラムを呼び出す場合、Natural プログラムでもデータベースコールを発行するのであれば、外部プログラムに独自の COMMIT ステートメントが含まれないようにする必要があります。呼び出し側の Natural プログラムが外部プログラムに代わって COMMIT ステートメントを発行する必要があります。

---

## 例

```
...  
DELETE FROM SQL-PERSONNEL WHERE NAME = 'SMITH'  
COMMIT  
...
```



# 144 DELETE - SQL

---

▪ 機能 .....	960
▪ 構文説明 .....	960

関連機能グループ：「[データベースへのアクセスと更新](#)」

『データベース管理システムインターフェイス』ドキュメントの次のセクションも参照してください。

- 「*Natural for DB2*」の「*NDB - DELETE - SQL*」
- 「*Natural for SQL/DS*」の「*DELETE*」

## 機能

---

SQL DELETE ステートメントは、カーソルを使用しないでテーブル内の行を削除したり（「[検索済](#) DELETE）、カーソルが位置づけられたテーブル内の行を削除したり（「[位置決め](#) DELETE）するために使用します。

## 構文説明


---

2つの異なる構造が可能です。

- [構文 1 - 検索済 DELETE](#)
- [構文 2 - 位置決め DELETE](#)

### 構文 1 - 検索済 DELETE

「検索済」DELETE ステートメントは、どの [SELECT](#) ステートメントにも関連しない独立したステートメントです。単一のステートメントで、ゼロ、1つ、複数、またはすべての行をテーブルから削除できます。削除される行はテーブルに適用する *search-condition* によって決定されます。オプションで、テーブル名に *correlation-name* を割り当てることができます。

 **注意:** 「検索済」DELETE で実際に削除された行の数は、システム変数 \*ROWCOUNT（『システム変数』ドキュメントを参照）で確認できます。

一般セットの構文：

```
DELETE FROM table-name [(correlation-name)] [WHERE search-condition]
```

拡張セットの構文：

```
DELETE FROM table-name [(correlation-name)] [WHERE search-condition]
    [ WITH { RR } { RS } { CS } ] [QUERYNO integer]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

構文要素の説明：

<b>FROM</b> <i>table-name</i>	<b>FROM 節：</b> この節では、行を削除するテーブルを指定します。
<i>correlation-name</i>	オプションで、テーブル名に <i>correlation-name</i> を割り当てることができます。
<b>WHERE</b> <i>search-condition</i>	<b>WHERE 節：</b> この節では、削除する行の選択条件を指定します。  WHERE 節を指定しないと、テーブル全体が削除されます。
<b>WITH</b>	<b>WITH 分離レベル節：</b> この節は <a href="#">SQL 拡張セット</a> に含まれます。  この節では、削除する行を検索するとき使用する分離レベルを明示的に指定できます。  詳細については、「 <a href="#">NDB - SELECT - Cursor-Oriented</a> 」の「 <a href="#">WITH - Isolation Level</a> 」（『データベース管理システムインターフェイス』ドキュメントの「 <a href="#">Natural for DB2</a> 」）を参照してください。
<b>QUERYNO</b> <i>integer</i>	<b>QUERYNO 節：</b> この節は <a href="#">SQL 拡張セット</a> に含まれます。  この節では、このステートメントの EXPLAIN 出力とトレースレコードで使用される番号を明示的に指定します。番号は、このステートメントの情報を含む行の PLAN_TABLET の QUERYNO 列として使用されます。

## 構文 2 - 位置決め DELETE

「位置決め」DELETE ステートメントは、常にデータベースループ内のカーソルを参照します。したがって、位置決めDELETE ステートメントで参照されるテーブルは、対応するSELECT ステートメントで参照されるテーブルと同一のものである必要があります。そうでない場合はエラーメッセージが返されます。位置決めDELETE を非カーソル選択で使用することはできません。

機能的に、位置決めDELETE ステートメントは「通常の」**Natural DELETE** ステートメントに対応します。

一般セットの構文：

```
DELETE FROM table-name WHERE CURRENT OF CURSOR [(r)]
```

拡張セットの構文：

```
DELETE FROM table-name WHERE CURRENT OF [ FOR ROW { [:]host-variable } OF ROWSET ] CURSOR [(r)]
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

構文要素の説明：

<b>FROM</b> <i>table-name</i> <b>WHERE CURRENT OF CURSOR</b>	<b>FROM 節：</b> この節では、行を削除するテーブルを指定します。
( <i>r</i> )	<b>ステートメント参照：</b> ( <i>r</i> ) 表記を使用して、削除する行の選択に使用されたステートメントを参照します。ステートメント参照の指定がない場合、DELETE ステートメントは、データベースレコードが選択された一番内側のアクティブな処理ループを参照します。
<b>FOR ROW ... OF ROWSET</b>	<b>FOR ROW ... OF ROWSET 節：</b> この節は <b>SQL 拡張セット</b> に含まれます。  位置決め SQL DELETE ステートメント用のオプションの FOR ROW ... OF ROWSET 節では、現在の行セットから削除する行を指定します。DELETE ステートメントが関連する <b>SELECT</b> ステートメントで行セットの位置決めを使用し、その INTO 節に列の配列が含まれる場合にのみ指定する必要があります。「 <a href="#">INTO 節</a> 」を参照してください。この節を指定しない場合、現在の行セットのすべての行が削除されます。

# 145

## INSERT - SQL

---

▪ 機能 .....	964
▪ 構文説明 .....	964
▪ 例 .....	969

一般セットの構文：

```
INSERT INTO table-name { (*)[VALUES-clause]
                        [(column-list)] VALUE-LIST }
```

拡張セットの構文：

```
(*) [OVERRIDING USER VALUE] VALUES(VIEW view-name)
INSERT INTO table-name { (*)[OVERRIDING USER VALUE][VALUES-clause]
                        [(column-list)] [OVERRIDING USER VALUE] }
                        VALUE-LIST
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[データベースへのアクセスと更新](#)」

『データベース管理システムインターフェイス』ドキュメントの次のセクションも参照してください。

- 「Natural for DB2」の「NDB - INSERT」
- 「Natural for SQL/DS」の「INSERT」

## 機能

SQL INSERT ステートメントは、1つ以上の行をテーブルに新しく追加するために使用します。

## 構文説明

<i>INTO table-name</i>	<p><b>INTO 節：</b></p> <p>INTO 節では、新しい行を挿入するテーブルを指定します。</p> <p>詳細については、<i>table-name</i> を参照してください。</p>
<i>column-list</i>	<p><b>列リスト：</b></p>

	<p>構文：</p> <pre>column-name...</pre> <p><i>column-list</i>には、1つ以上の <i>column-names</i> を指定できます。これは現在挿入される行内の値とともに指定します。</p> <p><i>column-list</i>を指定する場合、列の順序は <i>insert-item-list</i>に指定される値の順序または指定ビュー（下記参照）に含まれる値の順序と一致している必要があります。</p> <p><i>column-list</i>を指定しない場合、<i>insert-item-list</i>または指定ビュー内の値がすべての列の暗黙のリストに従ってテーブルに存在する順序で挿入されます。</p>
VALUES-clause	<p><b>VALUES 節：</b></p> <p>VALUES 節を使用して、1つの行をテーブルに挿入します。下記の「<a href="#">VALUES 節</a>」を参照してください。</p>
insert-item-list	<p><b>INSERT 1行：</b></p> <p><i>insert-item-list</i>には、<i>column-list</i>に指定された列に割り当てる1つ以上の値を指定できます。指定する値の順序が列の順序と一致している必要があります。</p> <p><i>column-list</i>を指定しない場合は、<i>insert-item-list</i>内の値がすべての列の暗黙のリストに従ってテーブルに存在する順序で挿入されます。</p> <p><i>insert-item-list</i>に指定できる値は、<i>constants</i>、<i>parameters</i>、<i>special-registers</i>、または NULL です。</p> <p><a href="#">view-name</a>、<a href="#">constant</a>、および <a href="#">parameter</a>については、「<a href="#">基本構文項目</a>」を参照してください。<a href="#">special-register</a>の情報も参照してください。</p> <p>値 NULL が割り当てられている場合は、指定されたフィールドが値（値 0 または「空白」でも）を受け取らないことを意味します。</p> <p>例 - INSERT 1行：</p> <pre>... INSERT INTO SQL-PERSONNEL (NAME,AGE) VALUES ('ADKINSON',35) ...</pre>
OVERRIDING USER VALUE	<p><b>OVERRIDING USER VALUE 節：</b></p> <p>この節は <a href="#">SQL 拡張セット</a>に含まれます。</p> <p>この節により、VALUES 節に指定された値、または GENERATED ALWAYS として定義されている列に対する全選択によって生成された値が無視されるようになります。</p>

## VALUES 節

VALUES 節を使用して、1つの行をテーブルに挿入します。アスタリスク (\*) が指定されているか、または *column-list* が指定されているかによって、VALUES 節の形式は次のいずれかになります。

### アスタリスク指定の VALUES 節

```
VALUES (VIEW view-name)
```

アスタリスク表記を指定する場合、VALUES 節にビューを指定する必要があります。このビューのフィールド値により、ビューのフィールド名を列名として使用して、新しい行が指定したテーブルに挿入されます。

### 列リスト指定の VALUES 節

```
[(column-list)] [OVERRIDING USER VALUE] VALUE-LIST
```

*column-list* を指定し、VALUES 節でビューを参照する場合、列リストに指定する項目数は、*VALUE-LIST* 内のビューに定義されたフィールド数に対応する必要があります。

*column-list* を指定しない場合は、ビューに定義されたフィールドが、すべての列の暗黙のリストに従って、指定したテーブルに存在する順序で挿入されます。

## VALUE-LIST

一般セットの構文：

```
{ VALUES { (VIEW view-name) } [FOR-n-ROWS-clause] }
```

拡張セットの構文：

```
{ VALUES { (VIEW view-name) } [FOR-n-ROWS-clause] }
{ [WITH_CTE common-table-expression,...] select-expression [ WITH { RR } [ [QUERYSNO ] ] ] } }
```



構文の説明：

<b>VIEW</b> <i>view-name</i>	このビューのフィールド値により、ビューのフィールド名を列名として使用して、新しい行が指定したテーブルに挿入されます。
<i>insert-item-list</i>	<p><b>INSERT 1 行：</b></p> <p><i>insert-item-list</i> には、列リストに指定された列に割り当てる 1 つ以上の値を指定できます。指定する値の順序が列の順序と一致している必要があります。</p> <p><i>column-list</i> を指定しない場合は、<i>insert-item-list</i> 内の値がすべての列の暗黙のリストに従ってテーブルに存在する順序で挿入されます。</p> <p><i>insert-item-list</i> に指定できる値は、定数、パラメータ、特別レジスタ、または NULL です。</p> <p><i>view-name</i>、<i>constant</i>、および <i>parameter</i> については、「基本構文項目」を参照してください。特別レジスタの情報も参照してください。</p> <p>値 NULL が割り当てられている場合は、指定されたフィールドが値（値 0 または空白でも）を受け取らないことを意味します。</p> <p>例 - INSERT 1 行：</p> <pre>... INSERT INTO SQL-PERSONNEL (NAME,AGE) VALUES ('ADKINSON',35) ...</pre>
<i>FOR-n-ROWS-clause</i>	オプションの節（下記参照）
<b>WITH_CTE</b> <i>common-table-expression</i>	<p>この節は <a href="#">SQL 拡張セット</a> に含まれます。</p> <p>このオプションの節では、後に続く SELECT ステートメントの任意の FROM 節で参照できる結果テーブルを定義できます。単一の WITH_CTE キーワードに続けて複数の共通テーブル式を指定できます。それぞれの共通テーブル式は、後続の共通テーブル式の FROM 節でも参照できます。</p> <p>詳細については、「<a href="#">SELECT - カーソル指向 ([WITH CTE common-table-expression,...] 構文)</a>」を参照してください。</p>
<i>select-expression</i>	<p>複数行の <b>INSERT</b>：</p> <p>この節は <a href="#">SQL 拡張セット</a> に含まれます。</p> <p><i>select-expression</i> では、<i>multiple</i> 行をテーブルに挿入します。<i>select-expression</i> が評価されると、結果テーブルの各行は、単一行の INSERT 処理の <a href="#">VALUES 節</a> の値として行に値が指定された場合と同様に扱われます。</p> <p>詳細については、「<a href="#">選択式</a>」を参照してください。</p>

	<p>例 - 複数行の INSERT :</p> <pre> ... INSERT INTO SQL-RETIREE (NAME,AGE,SEX)   SELECT LASTNAME, AGE, SEX   FROM SQL-EMPLOYEES   WHERE AGE &gt; 60 ... </pre> <p><b>注意:</b> 実際に挿入された行の数は、システム変数 *ROWCOUNT (『システム変数』ドキュメントを参照) で確認できます。</p>
<b>WITH RR/RS/CS</b>	<p><b>WITH</b> 分離レベル節 :</p> <p>この節は <b>SQL 拡張セット</b>に含まれます。</p> <p>この節では、挿入する行を検索するとき使用する分離レベルを明示的に指定できます。</p>
<b>QUERYNO_</b> <i>integer</i>	<p><b>QUERYNO</b> 節 :</p> <p>この節は <b>SQL 拡張セット</b>に含まれます。</p> <p>この節では、このステートメントの EXPLAIN 出力とトレース記録で使用される番号を明示的に指定します。</p>

### FOR-*n*-ROWS-Clause

```
FOR { [:]_host-variable } ROWS [ { ATOMIC
integer                                NOT ATOMIC CONTINUE ON SQLEXCEPTION } ]
```

この節は次の副節で構成されます。

#### FOR [:] *hostvariable/integer* ROWS 節

```
FOR { [:]_host-variable } ROWS
integer
```

この節の指定はオプションです。指定する必要があるのは、次の場合のみです。

- コンパイラオプション DB2ARRY を指定している。
- 複数の行が **VALUES 節**の *insert-item-list* で指定した配列から挿入される。

指定する場合、[:]*hostvariable/integer* では、**VALUES 節**の *insert-item-list* に指定した配列から DB2 テーブルに挿入する行の数を決定します (最初のオカレンスから挿入します)。

この節により、ループ内で Natural 配列から行を挿入するプログラムのパフォーマンスが向上します。この節を使用すると、配列内の行を 1 つの SQL ステートメントで挿入できます。

下記の例を参照してください。

『データベース管理システムインターフェイス』ドキュメントの *Natural for DB2* に関する説明も参照してください。

## ATOMIC 節

```
{ ATOMIC
  NOT ATOMIC CONTINUE ON SQLEXCEPTION }
```

この節では、複数の行の挿入を DB2 で原子処理として扱うかどうかを指定します。

指定する必要があるのは、次の場合のみです。

- コンパイラオプション DB2ARRY を指定している。
- 複数の行が **VALUES 節**の *insert-item-list* で指定した配列から挿入される。

構文の説明：

<b>ATOMIC</b>	エラーが発生した場合は、ターゲットテーブルに行を挿入しないことを指定します。これはデフォルト値です。
<b>NOT ATOMIC CONTINUE ON SQLEXCEPTION</b>	エラーが発生した場合は、エラーのないすべての行を挿入し、エラーのある行を DB2 で破棄することを指定します。

このような場合に返される SQL コードについては、DB2 SQL REFERENCE を参照してください。

## 例

```
DEFINE DATA LOCAL
01 NAME          (A20/1:10)  INIT <'ZILLER1','ZILLER2','ZILLER3','ZILLER4'
                                , 'ZILLER5','ZILLER6','ZILLER7','ZILLER8'
                                , 'ZILLER9','ZILLERA'>
01 ADDRESS       (A100/1:10) INIT <'ANGEL STREET 1','ANGEL STREET 2'
                                , 'ANGEL STREET 3','ANGEL STREET 4'
                                , 'ANGEL STREET 5','ANGEL STREET 6'
                                , 'ANGEL STREET 7','ANGEL STREET 8'
                                , 'ANGEL STREET 9','ANGEL STREET 10'>
01 DATENATD     (D/1:10)    INIT <D'1954-03-27',D'1954-03-27',D'1954-03-27'
                                ,D'1954-03-27',D'1954-03-27',D'1954-03-27'
                                ,D'1954-03-27',D'1954-03-27',D'1954-03-27'
                                ,D'1954-03-27'>
01 SALARY        (P4.2/1:10) INIT <1000,2000,3000,4000,5000
                                ,6000,7000,8000,9000,9999>
01 L$ADDRESS     (I2/1:10)  INIT <14,14,14,14,14,14,14,14,14,15>
```

## INSERT - SQL

---

```
01 N$ADDRESS (I2/1:10) INIT <00,00,00,00,00,00,00,00,00,00>
01 ROWS (I4)
01 INDEX (I4)
01 V1 VIEW OF NAT-DEMO_ID
02 NAME
02 ADDRESS (EM=X(20))
02 DATEOFBIRTH
02 SALARY
01 ROWCOUNT (I4)
END-DEFINE
OPTIONS DB2ARRY=ON /* <-- ENABLE DB2 ARRAY
ROWCOUNT := 10
INSERT INTO NAT-DEMO_ID
(NAME,ADDRESS,DATEOFBIRTH,SALARY)
VALUES
(:NAME(*), /* <-- ARRAY
:ADDRESS(*) /* <-- ARRAY
INDICATOR :N$ADDRESS(*) /* <-- ARRAY
LINDICATOR :L$ADDRESS(*), /* <-- ARRAY DB2 VCHAR
:DATENATD(1:10), /* <-- ARRAY NATURAL DATES
:SALARY(01:10) /* <-- ARRAY NATURAL PACKED
)
FOR :ROWCOUNT ROWS
SELECT * INTO VIEW V1 FROM NAT-DEMO_ID WHERE NAME > 'Z'
DISPLAY V1 /* <-- VERIFY INSERT
END-SELECT
END
```

# 146

## PROCESS SQL

---

■ 機能 .....	972
■ 構文説明 .....	972
■ 例 .....	973

```
PROCESS SQL ddm-name <<statement-string>>
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[データベースへのアクセスと更新](#)」

『データベース管理システムインターフェイス』ドキュメントの次のセクションも参照してください。

- 「Natural for DB2」の「NDB - PROCESS SQL」
- 「Natural for SQL/DS」の「PROCESS SQL」

## 機能

PROCESS SQL ステートメントは、基準データベースに対して SQL ステートメントを発行するために使用します。

## 構文説明

<i>ddm-name</i>	DDM の名前は、ストアードプロシージャを実行するデータベースの「アドレス」を示すように指定する必要があります。詳細については、「 <a href="#">ddm-name</a> 」を参照してください。
<i>statement-string</i>	<p><i>statement-string</i> に指定できるステートメントは、SQL ステートメント EXECUTE で発行できるステートメントと同じです（「<a href="#">フレキシブル SQL</a>」も参照）。</p> <p><b>注意:</b> Natural 環境と基準データベースのトランザクション同期の問題を避けるために COMMIT ステートメントおよび ROLLBACK ステートメントを PROCESS SQL 内で使用しないでください。</p> <p>ステートメント文字列は継続文字を指定しなくても複数のステートメント行にわたって指定できます。行の末尾にコメントを追加することも、行全体をコメント行にすることもできます。</p> <p>ステートメント文字列にはパラメータを含めることもできます。下記の「<a href="#">パラメータ</a>」を参照してください。</p>

## パラメータ

```
[ :U ] :host-variable [INDICATOR:host-variable] [LINIDICATOR:host-variable]
[ :G ]
```

説明した *parameter* と異なり、ここで述べる *host-variables* には先頭にコロン (:) を付ける必要があります。さらに、修飾子 (:U または :G) を付けることができます。

詳細については、*host-variable* を参照してください。

構文要素の説明：

<b>:U:host-variable</b>	接頭文字 :U は「Using」変数と呼ばれるホスト変数を修飾します。この変数は、値がデータベースに渡されることを示します。:U はデフォルト指定です。
<b>:G:host-variable</b>	接頭文字 :G は「Giving」変数と呼ばれるホスト変数を修飾します。この変数は、データベースから値を受け取ることを示します。

## 例

**DB2 の例 1 (z/OS) :**

```
PROCESS SQL DB2_DDM << CONNECT TO :LOCATION >>
```

**DB2 の例 2 (z/OS) :**

```
PROCESS SQL DB2_DDM << SET :G:LOCATION = CURRENT SERVER >>
```





# 147

## READ RESULT SET - SQL

---

▪ 機能 .....	976
▪ 制限事項 .....	976
▪ 構文説明 .....	977
▪ 例 .....	978

一般セットの構文：

```

READ [(limit)] RESULT SET result-set INTO { VIEW view-name } FROM ddm-name
      parameter
[GIVING [:] sql-code]
END-RESULT
    
```

拡張セットの構文：

```

READ [(limit)] RESULT SET result-set INTO { VIEW view-name } FROM ddm-name
      parameter
[WITH INSENSITIVE SCROLL [:] scroll-hv]
[GIVING [:] sql-code]
[WITH ROWSET POSITIONING FOR { [:] row_hv } ROWS]
      integer
END-RESULT
    
```

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[データベースへのアクセスと更新](#)」

『データベース管理システムインターフェイス』ドキュメントの「*Natural for DB2*」の「*NDB-READ RESULT SET*」も参照してください。

## 機能

SQL ステートメント READ RESULT SET は、CALLDBPROC ステートメントと組み合わせてのみ使用できます。前の CALLDBPROC ステートメントで呼び出されたストアードプロシージャによって作成された結果セットを読み込むために使用します。

## 制限事項

このステートメントは、Natural for SQL で使用することはできません。

## 構文説明

<i>limit</i>	読み込む行の数を制限できます。数値定数 (0~4294967295) または N、P または I フォーマットの変数として制限を指定できます。
<i>result-set</i>	結果セットとして、先行する <b>CALLDBPROC</b> ステートメントで代入された結果セットロケータ変数を指定します。結果セットには、フォーマット/長さ I4 の変数を指定する必要があります。  <b>注意:</b> CALLDBPROC ステートメントと READ RESULT SET ステートメント間の同期地点の処理が行われる場合、結果セットには READ RESULT SET ステートメントでアクセスできなくなります。
<b>FROM</b> <i>dgm-name</i>	<i>dgm-name</i> として、ストアードプロシージャを実行するデータベースの「アドレス指定」に使用する DDM の名前を指定します。詳細については、「 <i>dgm-name</i> 」を参照してください。
<b>WITH INSENSITIVE SCROLL [:]</b> <i>scroll_hv</i>	この節は <b>SQL 拡張セット</b> に含まれます。  この節を使用すると、アプリケーションはスクロール可能な固定カーソルを使用して、以前に呼び出したストアードプロシージャで作成された結果セットにアクセスします。この節を使用するためには、ストアードプロシージャによりスクロール可能なカーソル付きで結果セットが作成されている必要があります。 <i>scroll_hv</i> は、スクロール方向を含む英数字の Natural 変数である必要があります。 <i>scroll_hv</i> は、READ RESULT SET 処理ループが実行されるたびに評価されます。  <b>GIVING <i>sqlcode</i></b> オプションも指定した場合、RDBMS から <i>sqlcode</i> +100 (行は見つかりません) が返されたときでも、処理ループは開いています。  アプリケーションが ESCAPE ステートメントを発行した場合、または端末 I/O なしで 5 回連続して <i>sqlcode</i> +100 (行は見つかりません) になった場合に、処理は終了します。  <b>GIVING <i>sqlcode</i></b> オプションを指定しない場合は、処理ループは閉じられ、0 (成功) 以外の SQL コードがあれば RDBMS から返されます。
<b>GIVING</b> <i>sqlcode</i>	このオプションは、結果セットの処理に使用した SQL 「フェッチ」操作の SQL コードを取得するために使用できます。  このオプションを指定し、SQL 操作の SQL コードが 0 でない場合、Natural エラーメッセージは発行されません。この場合、SQL コード値に対して取られる動作は、呼び出し側の Natural オブジェクト内でコード化されている必要があります。  <i>sqlcode</i> フィールドには、フォーマット/長さ I4 の変数を指定する必要があります。  <b>GIVING <i>sqlcode</i></b> オプションを指定せず、SQL コードが 0 でない場合は、Natural エラーメッセージが発行されます
<b>WITH ROWSET POSITIONING FOR ... ROWS</b>	<b>WITH ROWSET POSITIONING FOR ... ROWS</b> 節： この節は <b>SQL 拡張セット</b> に含まれます。

	<p>この節を使用すると、アプリケーションはスクロール可能な固定カーソルを使用して、以前に呼び出したストアードプロシージャで作成された結果セットにアクセスします。この節を使用するためには、ストアードプロシージャによりスクロール可能なカーソル付きで結果セットが作成されている必要があります。 <i>scroll_hv</i> は、スクロール方向を含む英数字の <b>Natural</b> 変数である必要があります。 <i>scroll_hv</i> は、READ RESULT SET 処理ループが実行されるたびに評価されます。</p> <p>GIVING <i>sqlcode</i> オプションも指定した場合、RDBMS から <i>sqlcode</i>+100（行は見つかりません）が返されたときでも、処理ループは開いています。</p> <p>アプリケーションが ESCAPE ステートメントを発行した場合、または端末 I/O なしで 5 回連続して <i>sqlcode</i>+100（行は見つかりません）になった場合に、処理は終了します。</p> <p>GIVING <i>sqlcode</i> オプションを指定しない場合は、処理ループは閉じられ、0（成功）以外の SQL コードがあれば RDBMS から返されます。</p>
<b>END-RESULT</b>	<p>READ RESULT SET ステートメントを終了するには、<b>Natural</b> 予約キーワード <b>END-RESULT</b> を使用する必要があります。</p>

## 例

CALLDBPROC ステートメントの例を参照してください。

また、該当する **Natural** データベースインターフェイスのドキュメントを参照してください。

# 148

## ROLLBACK - SQL

---

▪ 機能 .....	980
▪ Natural 以外のプログラムについて .....	980
▪ 例 .....	981

## ROLLBACK

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[データベースへのアクセスと更新](#)」


『データベース管理システムインターフェイス』ドキュメントの次のセクションも参照してください。

- 「*Natural for DB2*」の「*NDB - ROLLBACK*」
- 「*Natural for SQL/DS*」の「*ROLLBACK*」

## 機能

SQL ROLLBACK ステートメントは *Natural BACKOUT TRANSACTION* ステートメントに対応します。最後のリカバリ単位の開始以降に行われたデータベース更新をすべて元に戻します。リカバリ単位は、セッションの開始後、または最後の SYNCPOINT、*COMMIT*、*END TRANSACTION*、または *BACKOUT TRANSACTION* ステートメントの後に開始できます。このステートメントも、トランザクション中に保持されていたすべてのレコードを解放します。

端末 I/O によりすでにコミットされている更新をプログラムでバックアウトしようとした場合、対応する *Natural* エラーメッセージ (NAT3711) が返されます。

-  **注意:** 作業の論理ユニット終了時には全カーソルがクローズされるため、ROLLBACK ステートメントはデータベース更新処理ループ内に指定しないでください。更新処理ループの外側、またはネストされたループの一番外側のループの後に指定する必要があります。

## Natural 以外のプログラムについて

*Natural* プログラムから他の標準プログラミング言語で記述された外部プログラムを呼び出す場合、*Natural* プログラムでもデータベースコールを発行するのであれば、外部プログラムに独自の ROLLBACK ステートメントが含まれないようにする必要があります。呼び出し側の *Natural* プログラムが外部プログラムに代わって ROLLBACK ステートメントを発行する必要があります。

## 例

---

```
...  
DELETE FROM SQL-PERSONNEL WHERE NAME = 'SMITH'  
ROLLBACK  
...
```





# 149 SELECT - SQL

---

▪ 機能 .....	984
▪ 構文説明 .....	984
▪ ジョインクエリ .....	997
▪ SELECT - カーソル指向 .....	997

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[データベースへのアクセスと更新](#)」

『データベース管理システムインターフェイス』ドキュメントの次のセクションも参照してください。

- 「Natural for DB2」の「*SELECT SINGLE - Non-Cursor-Oriented*」
- 「Natural for SQL/DS」の「*SELECT*」

## 機能

SELECT ステートメントでは、任意の行数を取得するためのカーソル選択と、多くても1つの行を取得する非カーソル選択（単独 SELECT）の両方がサポートされています。

SELECT ... END-SELECT 構造で、Natural は FIND ステートメントと同じデータベースループ処理を使用します。

## 構文説明

2つの異なる構造が可能です。

構文 1 - カーソル選択

一般セットの構文

SELECT <i>selection</i> INTO	{ <i>parameter</i> ,... VIEW { <i>view-name</i> [ <i>correlation-name</i> ]},... }	<i>table-expression</i>
[	{ UNION EXCEPT INTERSECT	[ALL] [(SELECT <i>selection table-expression</i> )]
ORDER BY	{ <i>integer</i> <i>column-reference</i> <i>expression</i>	[ ASC DESC ]
<i>statement</i> ...	{ END-SELECT (ストラク チャードモードのみ) LOOP (レポート モードのみ)	

拡張セットの構文：

[WITH_CTE <i>common-table-expression</i> ,...]			
SELECT <i>selection</i> INTO	{	<i>parameter</i> ,...	
		VIEW { <i>view-name</i>	
		[ <i>correlation-name</i> ]},...	
[	{	UNION	
		EXCEPT	
		INTERSECT	
		[ALL] [(SELECT <i>selection table-expression</i> )]	]
ORDER BY	{	<i>integer</i>	
		<i>column-reference</i>	
		<i>expression</i>	
		INPUT SEQUENCE	
		[ ASC ]	
		DESC ]	
[OPTIMIZE FOR <i>integer</i> ROWS]			
[	WITH	{	CS
			RR
			UR
			RS
			RS KEEP UPDATE LOCKS
			RR KEEP UPDATE LOCKS
		}	]
QUERYNO <i>integer</i>			
[	FETCH FIRST	{	1
			<i>integer</i>
		}	]
		{	ROW
			ROWS
		}	]
			ONLY ]
[WITH HOLD]			
[WITH RETURN]			
[	WITH	{	ASENSITIVE SCROLL
			INSENSITIVE SCROLL
			SENSITIVE STATIC
			SCROLL
			SENSITIVE DYNAMIC
			SCROLL
		}	]
			[:] <i>scroll_hv</i> [GIVING [:]
			<i>sqlcode</i>
[	WITH ROWSET	{	[:] <i>row_hv</i>
	POSITIONING		}
	FOR		ROWS ]
		<i>integer</i>	ROWS_RETURNED
			[:] <i>ret_row</i>
			]
[IF-NO-RECORDS-FOUND- <i>clause</i> ]			
<i>statement</i> ...			

END-SELECT (ストラク チャードモー ドのみ) LOOP (レポー ティングモー ドのみ)
--

## 構文要素の説明 - 構文 1 :

<b>SELECT</b> <i>selection</i>	<b>FIND</b> ステートメントと同様、カーソル選択は検索条件を基準に1つ以上のデータベーステーブルから行（レコード）のセットを選択するために使用します。また、アプリケーションプログラムからのカーソル管理は必要なく、Natural によって自動的に処理されます。詳細については、以下の「 <a href="#">SELECT - カーソル指向</a> 」を参照してください。
<b>INTO</b>	INTO 節は、選択の結果を格納するプログラム内のターゲットフィールドを指定するために使用します。詳細と例については、以下の「 <a href="#">INTO 節</a> 」を参照してください。
<b>VIEW</b>	INTO 節で1つまたは複数のビューを参照する場合は、 <i>selection</i> で指定した項目数が、ビューに定義されているフィールド（カウントするグループフィールドではなく、再定義フィールドとインジケータフィールド）の数と一致する必要があります。詳細と例については、以下の「 <a href="#">VIEW 節</a> 」を参照してください。
<i>table-expression</i>	<i>table-expression</i> は、FROM 節とオプションの WHERE 節から成ります。詳細と例については、以下の「 <a href="#">table-expression</a> 」を参照してください。
<b>UNION</b>	UNION は、複数の <i>select-expressions</i> の結果を結合します。詳細と例については、以下の「 <a href="#">UNION を伴うクエリ</a> 」を参照してください。
<b>ORDER BY</b>	ORDER BY 節は、SELECT ステートメントの結果を特定の順序で整列します。詳細と例については、以下の「 <a href="#">ORDER BY 節</a> 」を参照してください。
<b>IF NO RECORDS FOUND</b>	IF NO RECORDS FOUND 節は、先行する SELECT ステートメントで指定した選択条件に一致するレコードがない場合に処理ループを開始するために使用します。詳細については、以下の「 <a href="#">IF NO RECORDS FOUND 節</a> 」を参照してください。
<b>END-SELECT</b>	SELECT ステートメントを終了するには、Natural の予約キーワード END-SELECT を使用する必要があります。

次の構文要素は [SQL 拡張セット](#) に含まれます。

<b>WITH_CTE</b> <i>common-table-expression,...</i>	<b>WITH_CTE</b> <i>common-table-expression</i> : このオプションの節では、後に続く SELECT ステートメントの任意の FROM 節で参照できる結果テーブルを定義できます。単一の WITH_CTE キーワードに続けて複数の共通テーブル式を指定できます。それぞれの共通テーブル式は、後続の共通テーブル式の FROM 節でも参照できます。 詳細については、以下の「 <a href="#">SELECT - カーソル指向</a> 」を参照してください。
---	---

<b>OPTIMIZE FOR</b>	<p><b>OPTIMIZE FOR 節：</b></p> <p>詳細については、『データベース管理システムインターフェイス』ドキュメントの「<i>Natural for DB2</i>」を参照してください。</p>
<b>WITH CS/RS/UR/...</b>	<p><b>WITH CS/RS/UR/...節：</b></p> <p>この節では、ステートメントを実行するときの分離レベルを明示的に指定できます。この節の詳細については、『データベース管理システムインターフェイス』ドキュメントの「<i>Natural for DB2</i>」で「ステートメントとシステム変数」を参照してください。</p>
<b>QUERYNO</b>	<p><b>QUERYNO 節：</b></p> <p>QUERYNO 節では、この SQL ステートメントの EXPLAIN 出力とトレースレコードで使用される番号を指定します。この節の詳細については、『データベース管理システムインターフェイス』ドキュメントの「<i>Natural for DB2</i>」で「ステートメントとシステム変数」を参照してください。</p>
<b>FETCH FIRST</b>	<p><b>FETCH FIRST 節：</b></p> <p>この節では、フェッチできる行の数を制限します。この節の詳細については、『データベース管理システムインターフェイス』ドキュメントの「<i>Natural for DB2</i>」で「ステートメントとシステム変数」を参照してください。</p>
<b>WITH HOLD</b>	<p><b>WITH HOLD 節：</b></p> <p>詳細については、『データベース管理システムインターフェイス』ドキュメントの「<i>Natural for DB2</i>」で対応するセクションを参照してください。</p>
<b>WITH RETURN</b>	<p><b>WITH RETURN 節：</b></p> <p>詳細については、『データベース管理システムインターフェイス』ドキュメントの「<i>Natural for DB2</i>」で対応するセクションを参照してください。</p>
<b>WITH ... SCROLL</b>	<p><b>WITH ... SCROLL 節：</b></p> <p>この節では、DB2のスクロール可能カーソルを有効にします。スクロール可能カーソルには、ASENSITIVE、INSENSITIVE、SENSITIVE STATIC、または SENSITIVE DYNAMIC を使用できます。</p> <ul style="list-style-type: none"> <li>■ WITH ASENSITIVE SCROLL では、カーソルを INSENSITIVE または SENSITIVE DYNAMIC に指定します。どちらが使用されるかは、カーソルのオープン時にその読み取り専用プロパティに基づいて DB2 によって決定されます。カーソルが読み取り専用の場合、カーソルは INSENSITIVE になります。カーソルが読み取り専用でない場合、カーソルは SENSITIVE DYNAMIC になります。</li> <li>■ WITH INSENSITIVE SCROLL では、カーソルの更新後、基本テーブルを対象に実行される更新、削除、および挿入に対してカーソルを非センシティブにすることを指定します。INSENSITIVE SCROLL カーソルに対しては、位置決め UPDATE および位置決め DELETE は許可されていません。</li> </ul>

	<p>■ WITH SENSITIVE STATIC では、カーソルのオープン後、基本テーブルを対象に実行される更新および削除（挿入は含まない）に対してカーソルをセンシティブにすることを指定します。SENSITIVE STATIC SCROLL カーソルに対しては、位置決め UPDATE および位置決め DELETE が許可されています。</p> <p>■ WITH SENSITIVE DYNAMIC では、カーソルのオープン後、基本テーブルを対象に実行される更新、削除、および挿入に対してカーソルをセンシティブにすることを指定します。SENSITIVE DYNAMIC SCROLL カーソルに対しては、位置決め UPDATE および位置決め DELETE が許可されています。</p> <p>スクロール可能カーソルにより、カーソルがオープンされている限り、アプリケーションでカーソル位置の任意の行をいつでも位置決めできるようになります。</p> <p>位置決めは <code>scroll_hv</code> の内容に基づいて行われます。内容は、DB2 に対する FETCH が実行されるたびに評価されます。</p> <p>詳細については、「<a href="#">SELECT - カーソル指向</a>」を参照してください。</p>
--	---

## 構文 2 - 非カーソル選択

### 一般セットの構文

SELECT SINGLE	
<pre>selection INTO</pre>	<pre>{     parameter ,...     VIEW {view-name     [correlation-name ]}, ... }</pre>
} table-expression	
<pre>[IF-NO-RECORDS-FOUND-clause] statement...</pre>	
<pre>{ END-SELECT (structured mode only) }   LOOP (reporting mode only) }</pre>	

## 拡張セットの構文

SELECT SINGLE	
<pre>selection INTO</pre>	<pre>{   parameter ,...   VIEW {view-name   [correlation-name ]}, ... }</pre>
<pre>[   WITH {     CS     RR     UR   } ]</pre>	<pre>table-expression</pre>
<pre>[IF-NO-RECORDS-FOUND-clause] statement...</pre>	
<pre>{   END-SELECT (structured mode only)   LOOP (reporting mode only) }</pre>	

## 構文要素の説明 - 構文 2 :

<b>SELECT SINGLE</b>	SELECT SINGLE ステートメントでは、非カーソル選択（単独 SELECT）の機能、つまり、カーソルを使用せずに多くても 1 つの行を取得する選択式がサポートされています。これは <a href="#">位置決め UPDATE</a> や <a href="#">位置決め DELETE</a> ステートメントでは参照できません。詳細については、『データベース管理システムインターフェイス』ドキュメントの「Natural for DB2」の「SELECT SINGLE - Non-Cursor-Oriented」を参照してください。
<b>INTO</b>	INTO 節は、選択の結果を格納するプログラム内のターゲットフィールドを指定するために使用します。詳細と例については、以下の「 <a href="#">INTO 節</a> 」を参照してください。
<b>VIEW</b>	INTO 節で 1 つまたは複数のビューを参照する場合は、 <i>selection</i> で指定した項目数が、ビューに定義されているフィールド（カウントするグループフィールドではなく、再定義フィールドとインジケータフィールド）の数と一致する必要があります。詳細と例については、以下の「 <a href="#">VIEW 節</a> 」を参照してください。
<i>table-expression</i>	<i>table-expression</i> は、FROM 節とオプションの WHERE 節から成ります。詳細と例については、以下の「 <a href="#">table-expression</a> 」を参照してください。
<b>WITH CS/RR/UR</b>	<b>WITH CS/RR/UR 節 :</b>  この節では、ステートメントを実行するときの分離レベルを明示的に指定できます。この節の詳細については、『データベース管理システムインターフェイス』ドキュメントの「Natural for DB2」で「ステートメントとシステム変数」を参照してください。

<b>IF NO RECORDS FOUND</b>	IF NO RECORDS FOUND 節は、先行する SELECT ステートメントで指定した選択条件に一致するレコードがない場合に処理ループを開始するために使用します。詳細については、以下の「 <a href="#">IF NO RECORDS FOUND 節</a> 」を参照してください。
<b>END-SELECT</b>	SELECT ステートメントを終了するには、Natural の予約キーワード END-SELECT を使用する必要があります。

## INTO 節

```
INTO { parameter,...
      VIEW {view-name [correlation-name ]},... }
```

INTO 節は、選択の結果を格納するプログラム内のターゲットフィールドを指定するために使用します。INTO 節では、DEFINE DATA ステートメントで定義された個々の *parameters* または 1 つ以上のビューを指定できます。

全ターゲットフィールドは単一テーブルまたはジョイン処理でできた複数のテーブルから値を受け取ることができます（「[ジョインクエリ](#)」も参照）。



**注意:** 標準 SQL 構文では、INTO 節は非カーソル選択処理（単独 SELECT）でのみ使用されるため、単一行を選択する場合にのみ指定できます。ただし、Natural では、カーソル選択処理および非カーソル選択処理のどちらにも INTO 節を使用します。

*selection* はアスタリスク (\*) のみでも構成できます。標準選択式では、FROM 節に指定されたテーブル（複数可）の全列名のリストに対する省略形です。ただし、同じ構文項目 SELECT \* は、Natural SELECT ステートメントでは異なる意味になります。INTO 節にリストされた項目はすべて選択にも使用されます。それらの名前は既存のデータベースの列名に対応させる必要があります。

例：

例 1：

```
DEFINE DATA LOCAL
01 PERS VIEW OF SQL-PERSONNEL
   02 NAME
   02 AGE
END-DEFINE
...
SELECT *
  INTO NAME, AGE
```



例 2 :

```
...
SELECT *
  INTO VIEW PERS
```

上記の例は下記の例と同等です。

例 3 :

```
...
SELECT NAME, AGE
  INTO NAME, AGE
```


例 4 :

```
...
SELECT NAME, AGE
  INTO VIEW PERS
```

## VIEW 節

```
VIEW {view-name [correlation-name]}, ...
```

INTO 節で 1 つまたは複数のビューを参照する場合は、*selection* で指定した項目数が、ビューに定義されているフィールド（カウントするグループフィールドではなく、再定義フィールドとインジケータフィールド）の数と一致する必要があります。

 **注意:** Natural ターゲットフィールドおよびテーブルの列は両方とも Natural DDM に定義されている必要があります。ただし、割り当ては順序に従って行われるため、名前が異なってもかまいません。

ビュー使用の INTO 節の例 :

```
DEFINE DATA LOCAL
01 PERS VIEW OF SQL-PERSONNEL
   02 NAME
   02 AGE
END-DEFINE
...
SELECT FIRSTNAME, AGE
  INTO VIEW PERS
  FROM SQL-PERSONNEL
...
```

ターゲットフィールド `NAME` および `AGE` は `Natural` ビューの一部であり、テーブルの列 `FIRSTNAME` および `AGE` の内容を受け取ります。

<i>parameter</i>	<p>個々のパラメータをターゲットフィールドとして指定する場合、パラメータの数とフォーマットは、対応する選択（上記）に指定された <i>columns</i> または <i>scalar-expressions</i>（あるいは両方）の数とフォーマットに一致する必要があります（詳細については、「<a href="#">スカラー式</a>」を参照）。</p> <p>例：</p> <pre>DEFINE DATA LOCAL 01 #NAME   (A20) 01 #AGE    (I2) END-DEFINE ... SELECT NAME, AGE       INTO #NAME, #AGE       FROM SQL-PERSONNEL ...</pre> <p>ターゲットフィールド <code>#NAME</code> および <code>#AGE</code> は <code>Natural</code> プログラム変数であり、テーブルの列 <code>NAME</code> および <code>AGE</code> の内容を受け取ります。</p>
<i>correlation-name</i>	<p>複数のテーブルが連結（ジョイン）される <code>SELECT *</code> 構造で <code>VIEW</code> 節を使用する場合、指定したビューに含まれるフィールドが複数のテーブルに存在する列を参照するのであれば、<i>correlation-names</i>が必要となります。選択する列を識別するために、選択リスト生成時には、これらの列がすべて指定した <i>correlation-name</i> で修飾されます。ビューに割り当てる <i>correlation-name</i> は、連結（ジョイン）されるテーブルを修飾するために使用される <i>correlation-names</i> の1つに対応している必要があります。「<a href="#">ジョインクエリ</a>」も参照してください。</p> <p>例：</p> <pre>DEFINE DATA LOCAL 01 PERS VIEW OF SQL-PERSONNEL    02 NAME    02 FIRST-NAME    02 AGE END-DEFINE ... SELECT *       INTO VIEW PERS A       FROM SQL-PERSONNEL A, SQL-PERSONNEL B ...</pre>

## table-expression

*table-expression* は、FROM 節とオプションの WHERE 節から成ります。GROUP BY および HAVING 節は指定できません。

例 1 :

```
DEFINE DATA LOCAL
01 #NAME      (A20)
01 #FIRSTNAME (A15)
01 #AGE       (I2)
...
END-DEFINE
...
SELECT NAME, FIRSTNAME, AGE
  INTO #NAME, #FIRSTNAME, #AGE
  FROM SQL-PERSONNEL
  WHERE NAME IS NOT NULL
        AND AGE > 20
...
  DISPLAY #NAME #FIRSTNAME #AGE
END-SELECT
...
END
```

例 2 :

```
DEFINE DATA LOCAL
01 #COUNT    (I4)
...
END-DEFINE
...
SELECT SINGLE COUNT(*) INTO #COUNT FROM SQL-PERSONNEL
...
```

詳細については、[selection](#) および [table-expression](#) を参照してください。

## UNION を伴うクエリ



**注意:** 以降において、「SELECT ステートメント」は、UNION 演算と連結した複数の選択式で構成されるクエリ式全体の同義語として使用されます。

UNION は、複数の *select-expressions* の結果を結合します。各 *select-expressions* に指定された列は UNION 互換 (数、タイプ、フォーマットが一致) である必要があります。

UNION 演算子に ALL 修飾子が明示的に指定されていないと、重複した余分な行は常に UNION の結果から取り除かれます。ただし、UNION では、ALL に代わって明示的に指定できる DISTINCT オプションはありません。

例：

```

DEFINE DATA LOCAL
01 PERS VIEW OF SQL-PERSONNEL
  02 NAME
  02 AGE
  02 ADDRESS (1:6)
END-DEFINE
...
SELECT NAME, AGE, ADDRESS
  INTO VIEW PERS
  FROM SQL-PERSONNEL
  WHERE AGE > 55
UNION ALL
SELECT NAME, AGE, ADDRESS
  FROM SQL-EMPLOYEES
  WHERE PERSNR < 100
ORDER BY NAME
...
END-SELECT
...

```

一般的に、任意の数の *select-expressions* を UNION で連結できます。

INTO 節は、最初の *select-expression* にのみ指定する必要があります。

ORDER BY 節は、最後の *select-expression* の後に指定する必要があります。並べ替える列は名前ではなく番号で識別する必要があります。

## ORDER BY 節

```

ORDER BY { { integer } [ ASC ] } { column-reference } [ DESC ] } ,...

```

ORDER BY 節は、SELECT ステートメントの結果を特定の順序で整列します。

各 ORDER BY 節には、結果テーブルの列を指定する必要があります。ほとんどの ORDER BY 節で、*column-reference* (オプションで修飾された列名)、または列番号によって列を識別できます。UNION を指定したクエリでは、列は列番号で識別する必要があります。列番号は、*selection* 内で左から右へ位置づけ、*integer* 値で示します。この機能により、名前を持たない計算列に基づいて結果を整列できるようになります。

例：

```
DEFINE DATA LOCAL
1 #NAME          (A20)
1 #YEARS-TO-WORK (I2)
END-DEFINE
...
SELECT NAME , 65 - AGE
      INTO #NAME, #YEARS-TO-WORK
      FROM SQL-PERSONNEL
      ORDER BY 2
...

```

ORDER BY 節に指定する順序は昇順 (ASC) または降順 (DESC) です。ASC がデフォルトです。

例：

```
DEFINE DATA LOCAL
1 PERS VIEW OF SQL-PERSONNEL
1 NAME
1 AGE
1 ADDRESS      (1:6)
END-DEFINE
...
SELECT NAME, AGE, ADDRESS
      INTO VIEW PERS
      FROM SQL-PERSONNEL
      WHERE AGE = 55
      ORDER BY NAME DESC
...

```

詳細については、[integer](#) および [column-reference](#) を参照してください。

## IF NO RECORDS FOUND 節



**注意:** この節は実際には Natural SQL に属するものではありません。これは Natural 機能が SQL ループ処理に対して有効にされたことを表します。

ストラクチャードモード構文

```
IF NO [RECORDS] [FOUND]
{
    ENTER
    statement...
}
END- NOREC
```


## レポーティングモード構文

```
IF NO [RECORDS] [FOUND]
{
    ENTER
    statement
    DO statement... DOEND
}
```

IF NO RECORDS FOUND 節は、先行する SELECT ステートメントで指定した選択条件に一致するレコードがない場合に処理ループを開始するために使用します。

指定した選択条件に一致するレコードがない場合、IF NO RECORDS FOUND 節により、「空」のレコードを使用して処理ループが1回実行されます。これが望ましくない場合は、IF NO RECORDS FOUND 節に ESCAPE BOTTOM ステートメントを指定します。

IF NO RECORDS FOUND 節に1つ以上のステートメントを指定すると、処理ループに入る直前にそのステートメントが実行されます。ループに入る前に実行するステートメントがない場合は、キーワード ENTER を使用する必要があります。

 **注意:** SELECT ステートメントの結果セットが単一行の NULL 値で構成される場合、IF NO RECORDS FOUND 節は実行されません。この状況は、選択リストが列に対する集積関数 SUM、AVG、MIN、MAX の1つで単独に構成され、集積関数で使用するセットが空の場合に起こることがあります。このような方法で集積関数を使用するときは、IF NO RECORDS FOUND 節を使用せずに、対応する空値インジケータフィールドの値をチェックする必要があります。

## データベース値

IF NO RECORDS FOUND 節内のステートメントで他の値の割り当てが行われない限り、Natural は現在のループで指定されたファイルを参照するすべてのデータベースフィールドをリセットして空にします。

## システム機能の評価

Natural システム機能は、IF NO RECORDS FOUND 節の結果として処理用に作成される空のレコードに対して1回評価されます。

## ジョインクエリ

ジョインとは複数のテーブルからデータを取得するクエリです。参照されるテーブルはすべて FROM 節に指定する必要があります。

例：

```
DEFINE DATA LOCAL
1 #NAME      (A20)
1 #MONEY     (I4)
END-DEFINE
...
SELECT NAME, ACCOUNT
INTO #NAME, #MONEY
FROM SQL-PERSONNEL P, SQL-FINANCE F
WHERE P.PERSNR = F.PERSNR
AND F.ACCOUNT > 10000
...
```

ジョインは常に FROM 節にリストされたテーブルのデカルト積を形成し、このデカルト積から WHERE 節に指定されたジョイン条件を満たさない行をすべて削除します。

テーブル名が長すぎる場合、*Correlation-names* を使用して書き込みを保存できます。選択リストに指定された列が、連結（ジョイン）される複数のテーブルに存在するとき、同じ名前のどの列を選択するかを識別するために *Correlation-names* を使用する必要があります。

## SELECT - カーソル指向

Natural **FIND** ステートメントと同様、カーソル指向の SELECT ステートメントは検索条件を基準に 1 つ以上の DB2 テーブルから行（レコード）のセットを選択するために使用します。データベースループが開始されるため、ループは LOOP（レポーティングモード）または END-SELECT ステートメントで閉じる必要があります。この構造では、Natural は FIND ステートメントの場合と同じ処理ループを使用します。

また、アプリケーションプログラムからのカーソル管理は必要なく、Natural によって自動的に処理されます。

以下に参考情報を示します。

- [WITH\\_CTE common-table-expression,...](#)
- [OPTIMIZE FOR integer ROWS](#)
- [WITH - 分離レベル](#)
- [QUERYNO](#)
- [FETCH FIRST](#)

- WITH HOLD
- WITH RETURN
- WITH INSENSITIVE/SENSITIVE

### WITH\_CTE common-table-expression,...

この節では、後に続く SELECT ステートメントの任意の FROM 節で参照できる結果テーブルを定義できます。

Natural 固有のキーワード WITH\_CTE は、SQL キーワード WITH に対応します。WITH\_CTE は、Natural コンパイラにより SQL キーワード WITH に変換されます。

共通テーブル式はいずれも、次の構文に従っている必要があります。

```
[common-table-expression-name [(column-name,...)] AS (fullselect)]
```

構文の説明：

<i>common-table-expression-name</i>	修飾されていない SQL 識別子である必要があります。また、同じステートメント内で指定されている他の共通テーブル式とは異なっている必要があります。  <i>common-table-expression-name</i> はいずれも、後に続く任意の <i>common-table-expression-name</i> の FROM 節、または後に続く SELECT ステートメントの FROM 節の中で指定できます。
<i>column-name</i>	修飾されていない SQL 識別子である必要があります。また、1つの <i>common-table-expression-name</i> 内で一意である必要があります。
AS ( <i>fullselect</i> )	<i>column-names</i> の数は、 <i>fullselect</i> の列の数と一致している必要があります。

共通テーブル式は、次のような場合に使用できます。

- ビューの作成を回避するため、ビューの代わりとして使用する場合
- *fullselect* 内で同じ結果テーブルを共有する必要がある場合
- 反復を使用して結果を導出する必要がある場合

反復を使用するクエリは、材料の請求書などの用途に便利です。



例：

```
WITH_CTE
RPL (PART,SUBPART,QUANTITY) AS
(SELECT ROOT.PART,ROOT.SUBPART,ROOT.QUANTITY
 FROM HGK-PARTLIST ROOT
 WHERE ROOT.PART = '01'
 UNION ALL
 SELECT CHILD.PART,CHILD.SUBPART,CHILD.QUANTITY
 FROM RPL PARENT, HGK-PARTLIST CHILD
 WHERE PARENT.SUBPART = CHILD.PART
 )
SELECT DISTINCT PART,SUBPART,QUANTITY
 INTO VIEW V1
 FROM RPL
 ORDER BY PART,SUBPART,QUANTITY
END-SELECT
```

## OPTIMIZE FOR integer ROWS

[OPTIMIZE FOR *integer*ROWS]

OPTIMIZE FOR *integer* ROWS 節は、結果テーブルから取得する行数 (*integer*) を前もって DB2 に知らせるために使用します。この節を使用しない場合、DB2 では結果テーブルのすべての行が取得されると仮定し、それに応じて最適化を行います。

このオプションの節は、選択される行数を把握している場合に役立ちます。実際に選択される行数が *integer* 値 (0~2147483647) を超えない場合、*integer* の行を最適化することでパフォーマンスが向上します。

例：

```
SELECT name INTO
#name FROM table WHERE AGE = 2 OPTIMIZE FOR 100 ROWS
```

## WITH - 分離レベル

WITH	}	CS	}
		RR	
		RR KEEP UPDATE LOCK	
		RS	
		RS KEEP UPDATE LOCKS	
		UR	

このWITH節では、ステートメントを実行するときの分離レベルを明示的に指定できます。次のオプションを使用できます。

オプション	意味
CS	カーソルの安定
RR	繰り返し可能な読み込み
RS	読み込みの安定
RS KEEP UPDATE LOCKS	FOR UPDATE OF 節を指定している場合にのみ有効です。 読み込みの安定と更新ロックの保持です。
RR KEEP UPDATE LOCKS	FOR UPDATE OF 節を指定している場合にのみ有効です。 繰り返し可能な読み込みと更新ロックの保持です。
UR	未コミットの読み込み

WITH UR は、テーブルが読み取り専用のときに SELECT ステートメント内にのみ指定できます。デフォルトの分離レベルは、ステートメントがバインドされるパッケージまたはプランの分離によって決定されます。また、デフォルトの分離レベルは、結果テーブルが読み取り専用かどうかによっても変わります。デフォルトの分離レベルについては、IBM 文献を参照してください。



**注意:** また、このオプションは非カーソル選択にのみ有効です。

## QUERYNO

```
[QUERYNO integer]
```

QUERYNO 節では、この SQL ステートメントの EXPLAIN 出力とトレースレコードで使用される番号を指定します。番号は、このステートメントの情報を含む行の PLAN\_TABLET の QUERYNO 列として使用されます。

## FETCH FIRST

```
[ FETCH FIRST {  $\frac{1}{integer}$  } { ROWS } ONLY ]
```

FETCH FIRST 節では、フェッチする行の数を制限します。ごく限られた数の行が必要な場合にこの節を使用すると、結果セットの規模が大きくなる可能性のあるクエリのパフォーマンスが向上します。

## WITH HOLD

[WITH HOLD]

WITH HOLD 節は、データベースループ内のコミット処理によってカーソルがクローズされないようにするために使用します。WITH HOLD を指定した場合、コミット処理により現在の作業論理ユニットの更新はすべてコミットされますが、この際、カーソルを維持する必要のないロックしか解放されません。この節は、主にバッチモードで役立ちます。CICS 擬似会話型モードおよび IMS メッセージドリブンプログラムでは無視されます。

例：

```
SELECT name INTO #name FROM table
WHERE AGE = 2 WITH HOLD
```

## WITH RETURN

[WITH RETURN]

WITH RETURN 節は、結果セットを作成するために使用します。したがって、この節は Natural スタアドプロシージャとして動作するプログラムにのみ適用されます。SELECT ステートメントに WITH RETURN 節を指定した場合、関連する処理ループを抜けたときに基準となるカーソルがオープンのまま維持されます。ただし、処理ループで結果セット自体のすべての行が読み込まれた場合を除きます。処理ループの最初の実行時には、カーソルのみがオープンされます。最初の行はまだフェッチされません。これにより、Natural プログラムはスタアドプロシージャの呼び出し元に完全な結果セットを返すことができます。Natural スタアドプロシージャで処理する行の数、およびスタアドプロシージャの呼び出し元に返す結果セットの未処理の行の数は自由に決めることができます。Natural スタアドプロシージャで選択処理の行を処理する場合、次を定義する必要があります。

```
IF *counter =1 ESCAPE TOP END-IF
```

これにより、処理ループで最初の「空の行」が処理されることがなくなります。行の処理を終了する場合、次を定義する必要があります。

```
If condition ESCAPE BOTTOM END-IF
```

この定義は処理ループで行います。

プログラムで結果セットのすべての行を読み込む場合、カーソルはクローズされ、この SELECT WITH RETURN に対する結果セットはスタアドプロシージャの呼び出し元には返されません。

次にプログラムの例を示します。例 1 では完全な結果セットを取得し、例 2 では部分的な結果セットを取得しています。

## 例 1 :

```

DEFINE DATA LOCAL
. . .
END DEFINE
*
* Return all rows of the result set
*
SELECT * INTO VIEW V2
        FROM SYSIBM-SYSROUTINES
        WHERE RESULT_SETS > 0
        WITH RETURN

ESCAPE BOTTOM
END-SELECT
END

```

## 例 2 :

```

DEFINE DATA LOCAL
. . .
END DEFINE
*
* Read the first two rows and return the rest as result set
*
SELECT * INTO VIEW V2
        FROM SYSIBM-SYSROUTINES
        WHERE RESULT_SETS > 0
        WITH RETURN

WRITE PROCEDURE *COUNTER
IF *COUNTER = 1 ESCAPE TOP END-IF
IF *COUNTER = 3 ESCAPE BOTTOM END-IF
END-SELECT
END

```

**WITH INSENSITIVE/SENSITIVE**

<pre> WITH {   ASENSITIVE SCROLL   INSENSITIVE SCROLL   SENSITIVE STATIC SCROLL   SENSITIVE DYNAMIC SCROLL } [:] <i>scroll_hv</i> [GIVING [:] <i>sqlcode</i>] </pre>
--

Natural for DB2 では、DB2 のスクロール可能カーソルを利用できます。そのためには、WITH ASENSITIVE SCROLL 節、WITH SENSITIVE STATIC SCROLL 節、および SENSITIVE DYNAMIC SCROLL 節を使用します。スクロール可能カーソルにより、Natural for DB2 アプリケーションで結果セット内の任意の行をランダムに位置決めできるようになります。スクロール不可能なカーソルでは、データは上から下に順番に読み込むことしかできません。

ASENSITIVE スクロール可能カーソルは、INSENSITIVE (カーソルが READ-ONLY の場合) または SENSITIVE DYNAMIC (カーソルが READ-ONLY でない場合) です。

INSENSITIVE および SENSITIVE STATIC スクロール可能カーソルでは一時的な結果テーブルを使用し、DB2 に TEMP データベースが必要になります（IBM の関連 DB2 文献を参照）。

INSENSITIVE SCROLL は、位置決め UPDATE または位置決め DELETE 処理に使用できないカーソルを表します。また、INSENSITIVE SCROLL カーソルは、一度オープンされると、その後に基本テーブルを対象に実行される UPDATE、DELETE、または INSERT を反映しません。

SENSITIVE STATIC SCROLL は、位置決め UPDATE または位置決め DELETE 処理に使用できるカーソルを表します。また、SENSITIVE STATIC SCROLL カーソルは、基本テーブルの行の UPDATE、DELETE を反映します。ただし、INSERT 処理は反映しません。

SENSITIVE DYNAMIC スクロール可能カーソルは、カーソルがオープンされている間、基本テーブルを対象に実行される UPDATE、DELETE、および INSERT を反映します。

以下に参考情報を示します。

- [scroll\\_hv](#)
- [scroll\\_hv - 感度指定](#)
- [scroll\\_hv - オプション](#)
- [GIVING \[:\] sqlcode](#)

### scroll\_hv

変数 `scroll_hv` は英数字にする必要があります。

変数 `scroll_hv` では、データベース処理ループの 1 回の実行でフェッチする結果テーブルの行を指定します。また、FETCH 処理中に基本テーブルを対象に実行される UPDATE または DELETE に対する感度を指定します。`scroll_hv` の内容は、データベース処理ループサイクルが実行されるたびに評価されます。

$\left[ \left\{ \begin{array}{l} \underline{I}NSENSITIVE \\ \underline{S}ENSITIVE \end{array} \right\} \right]$	$\left\{ \begin{array}{l} \underline{A}FTER \\ \underline{B}EFORE \\ \underline{C}URRENT \\ \underline{F}IRST \\ \underline{L}AST \\ \underline{P}RIOR \\ \underline{N}EXT \mid N \\ \left\{ \begin{array}{l} \underline{A}BSOLUTE \\ \underline{R}ELATIVE \end{array} \right\} \quad [+ -] \textit{integer} \end{array} \right\}$
---	--

**scroll\_hv - 感度指定**

感度 INSENSITIVE または SENSITIVE の指定はオプションです。

INSENSITIVE SCROLL カーソルに対する FETCH から省略した場合、デフォルトで INSENSITIVE が使用されます。

SENSITIVE STATIC/DYNAMIC SCROLL カーソルに対する FETCH から省略した場合、デフォルトで SENSITIVE が使用されます。

感度では、スクロール可能カーソルに対する FETCH 処理の実行時に基本テーブル内の行をチェックするかどうかを指定します。

対応する基本テーブルの列が WHERE 節の対象であり、削除されなかった場合、SENSITIVE FETCH によって基本テーブルの行が返されます。

対応する基本テーブルの列が WHERE 節の対象でない場合、または削除されなかった場合、SENSITIVE FETCH によって UPDATE ホール状態または DELETE ホール状態 (SQLCODE +222) が返されます。

INSENSITIVE FETCH では、対応する基本テーブルの列はチェックされません。

**scroll\_hv - オプション**

フェッチする行、フェッチの開始位置、およびスクロール方向を決定するためのオプションについて次に説明します。

オプション	説明
AFTER	最後の行の後に位置決めします。行はフェッチされません。
BEFORE	最初の行の前に位置決めします。行はフェッチされません。
CURRENT	現在の行を (再度) フェッチします。
FIRST	最初の行をフェッチします。
LAST	最後の行をフェッチします。
NEXT	現在の行の次の行をフェッチします。これはデフォルト値です。
PRIOR	現在の行の前の行をフェッチします。
+/- integer	ABSOLUTE または RELATIVE との接続でのみ適用されます。 フェッチする行の位置を ABSOLUTE または RELATIVE で指定します。 整数の前にプラス記号 (+) またはマイナス記号 (-) を入力します。 デフォルト値はプラス (+) です。
ABSOLUTE	+/- integer との接続でのみ適用されます。 行をフェッチする結果セット内の絶対位置として integer を使用します。

オプション	説明
	正または負の位置番号の詳細については、IBM 提供の DB2 SQL リファレンスを参照してください。
RELATIVE	+/- <i>integer</i> との接続でのみ適用されます。 行をフェッチする結果セット内の現在の位置への相対位置として <i>integer</i> を使用します。 正または負の位置番号の詳細については、IBM 提供の DB2 SQL リファレンスを参照してください。

### GIVING [:] sqlcode

GIVING [:] *sqlcode* の指定はオプションです。指定する場合、Natural 変数 [:] *sqlcode* のフォーマットは I4 にする必要があります。この変数の値は、基準となる FETCH 処理の DB2 SQLCODE から返されます。これにより、スクロール可能カーソルがオープンされている間に発生した各種ステータスにアプリケーションで対応できるようになります。SQLCODE で示される最も重要なステータスコードを次の表に示します。

SQLCODE	説明
0	FETCH 処理は正常に完了しました。データが返されました（ただし、オプション BEFORE または AFTER を指定した FETCH を除く）。
+100	行は見つかりません。カーソルはオープンされたままです。データは返されませんでした。
+222	UPDATE または DELETE ホールです。カーソルはオープンされたままです。データは返されませんでした。基本テーブルの対応する行が更新または削除されました。そのため、行は WHERE 節の対象ではなくなりました。
+231	オプション CURRENT を指定したフェッチ処理ですが、カーソルが行に位置づけられていません。データは返されませんでした。これは前の FETCH によって SQLCODE +100 が返された場合に発生します。

GIVING [:] *sqlcode* を指定する場合、アプリケーションで各種ステータスに対応できる必要があります。端末 I/O なしで 5 回連続して SQLCODE +100 になった場合、アプリケーションループを回避するために、NDB ランタイムから Natural エラー NAT3296 が発行されます。アプリケーションでは、ESCAPE ステートメントを実行して処理ループを終了できます。

GIVING [:] *sqlcode* を指定しない場合、SQLCODE 0 および SQLCODE +100 を除き、各 SQLCODE によって Natural エラー NAT3700 が生成され、処理ループが終了します。SQLCODE +100（行は見つかりません）では処理ループが終了します。

Natural システムライブラリ SYSDB2 内のプログラム例 DEM2SCRL も参照してください。





# 150 UPDATE - SQL

---

▪ 機能 .....	1008
▪ 構文説明 .....	1008
▪ 例 .....	1011

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

関連機能グループ：「[データベースへのアクセスと更新](#)」

『データベース管理システムインターフェイス』ドキュメントの次のセクションも参照してください。

- 「*Natural for DB2*」の「*NDB - UPDATE - SQL*」
- 「*Natural for SQL/DS*」の「*UPDATE*」

## 機能

SQL UPDATE ステートメントは、カーソルを使用しないでテーブル内の行に対して UPDATE（更新）処理を実行したり（「[検索済](#) UPDATE）、カーソルが位置づけられた行の列に対して更新処理を実行したり（「[位置決め](#) UPDATE）するために使用します。


## 構文説明

2つの異なる構造が可能です。

- [構文 1 - 検索済 UPDATE](#)
- [構文 2 - 位置決め UPDATE](#)

### 構文 1 - 検索済 UPDATE

「検索済」 UPDATE ステートメントは、どの SELECT ステートメントにも関連しない独立したステートメントです。単一のステートメントで、テーブルのゼロ、1つ、複数、またはすべての行を更新できます。更新される行はテーブルに適用する *search-condition* によって決定されます。オプションで、ビュー名およびテーブル名に *correlation-name* を割り当てることができます。

 **注意:** 「検索済」 UPDATE で実際に更新された行の数は、システム変数 \*ROWCOUNT（『システム変数』ドキュメントを参照）で確認できます。

UPDATE	{	<i>view-name</i> [ <i>correlation-name</i> ] SET *	}
		<i>table-name</i> [ <i>correlation-name</i> ] SET <i>assignment-list</i>	
[WHERE <i>search-condition</i> ]	[	WITH {	[QUERY NO <i>integer</i> ]
		RR RS CS }	
	]		

構文図で使用されている記号については、「[構文記号](#)」を参照してください。

## 構文要素の説明 - 構文 1 :

<i>view-name</i>	ビュー名： DEFINE DATA ステートメントでの定義に従って Natural ビューの名前を参照します。詳細については、「 <a href="#">基本構文項目</a> 」の <i>view-name</i> を参照してください。
<b>SET</b>	<b>SET 節：</b> ビューに更新を指定している場合は、ビューのすべての列が更新される必要があるため、SET 節にアスタリスク (*) を指定する必要があります。 テーブルに更新を指定している場合は、SET 節には <i>assignment-list</i> または更新する列を含むビューの名前を指定する必要があります。
<i>assignment-list</i>	下記の「 <a href="#">割り当てリスト</a> 」を参照してください。
<b>WHERE</b> <i>search-condition</i>	<b>WHERE 節：</b> この節では、更新する行の選択条件を指定します。 WHERE 節を指定しないと、テーブル全体が更新されます。

## 割り当てリスト

$$\left\{ \text{column-name} = \left\{ \text{scalar-expression NULL} \right\} \right\} \dots$$

*assignment-list* では、1つ以上の列に値を割り当てることができます。値は *scalar-expression* または NULL のいずれかです。詳細については、「[スカラー式](#)」を参照してください。

値 NULL が割り当てられている場合は、指定されたフィールドに値（値 0 または「空白」でも）は含まれないことを意味します。

## SQL 拡張セット

次の構文要素は [SQL 拡張セット](#) に含まれます。

<i>correlation-name</i>	項目 <i>correlation-name</i> は、 <i>table-name</i> のエイリアス名を表します。詳細については、「 <a href="#">基本構文項目</a> 」の <i>correlation-name</i> を参照してください。
<b>WITH</b>	<b>WITH 分離レベル節：</b> この節では、更新する行を検索するときに使用する分離レベルを明示的に指定できます。詳細については、「 <a href="#">NDB - SELECT - Cursor-Oriented</a> 」の「 <a href="#">WITH - Isolation Level</a> 」（『データベース管理システムインターフェイス』ドキュメントの「 <a href="#">Natural for DB2</a> 」）を参照してください。
<b>QUERYNO</b> <i>integer</i>	<b>QUERYNO 節：</b>

この節では、このステートメントの EXPLAIN 出力とトレースレコードで使用される番号を明示的に指定できます。番号は、このステートメントの情報を含む行の PLAN\_TABLET の QUERYNO 列として使用されます。

## 構文 2 - 位置決め UPDATE

「位置決め」 UPDATE ステートメントは、常にデータベースループ内のカーソルを参照します。したがって、位置決め UPDATE ステートメントで参照されるテーブルまたはビューは、対応する SELECT ステートメントで参照されるテーブルまたはビューと同一のものである必要があります。そうでない場合はエラーメッセージが返されます。位置決め UPDATE を非カーソル選択で使用することはできません。

一般セットの構文

```
UPDATE { view-name SET *
        view-name SET assignment-list } [WHERE CURRENT OF CURSOR (r)]
```

拡張セットの構文

```
UPDATE { view-name SET *
        view-name SET
        assignment-list } [WHERE CURRENT OF [ FOR
        ROW { [:]host-variable } OF ROWSET ]
```

構文要素の説明 - 構文 2 :

<i>view-name</i>	ビュー名： DEFINE DATA ステートメントでの定義に従って Natural ビューの名前を参照します。詳細については、「基本構文項目」の <i>view-name</i> を参照してください。
<b>SET *</b> <b>SET</b> <i>assignment-list</i>	<b>SET 節：</b> ビューに更新を指定している場合は、ビューのすべての列が更新される必要があるため、SET 節にアスタリスク (*) を指定する必要があります。 テーブルに更新を指定している場合は、SET 節には <i>assignment-list</i> または更新する列を含むビューの名前を指定する必要があります。
<b>WHERE CURRENT OF</b> <b>CURSOR (r)</b>	ステートメント参照： (r) 表記を使用して、更新する行の選択に使用されたステートメントを参照します。ステートメント参照の指定がない場合、UPDATE ステートメントは、データベースレコードが選択された一番内側のアクティブな処理ループを参照します。
<b>FOR ROW ... OF</b> <b>ROWSET</b>	<b>FOR ROW ... OF ROWSET 節：</b> この節は <a href="#">SQL 拡張セット</a> に含まれます。

	<p>位置決め SQL UPDATE ステートメント用のオプションの FOR ROW ... OF ROWSET 節では、現在の行セットで更新する行を指定します。UPDATE ステートメントが関連する SELECT ステートメントで行セットの位置決めを使用し、その <b>INTO</b> 節に列の配列が含まれる場合にのみ指定する必要があります。</p> <p>この節を指定しない場合、現在の行セットのすべての行が <i>assignment-list</i> 内の値で更新されます。</p> <p><i>view-name</i> SET * を指定している場合、この節は指定できません。</p>
--	---

## 例

- 例 1 - 検索済 UPDATE
- 例 2 - *assignment-list* を使用した検索済 UPDATE
- 例 3 - 位置決め UPDATE
- 例 4 - *assignment-list* を使用した位置決め UPDATE

### 例 1 - 検索済 UPDATE

```

DEFINE DATA LOCAL
1 PERS VIEW OF SQL-PERSONNEL
2 NAME
2 AGE
...
END-DEFINE
...
ASSIGN AGE = 45
ASSIGN NAME = 'SCHMIDT'
UPDATE PERS SET * WHERE NAME = 'SCHMIDT'
...

```

### 例 2 - *assignment-list* を使用した検索済 UPDATE

```

DEFINE DATA LOCAL
1 PERS VIEW OF SQL-PERSONNEL
2 NAME
2 AGE
...
END-DEFINE
...
UPDATE SQL-PERSONNEL SET AGE = AGE + 1 WHERE NAME = 'SCHMIDT'
...

```

**例 3 - 位置決め UPDATE**

```
DEFINE DATA LOCAL
1 PERS VIEW OF SQL-PERSONNEL
2 NAME
2 AGE
...
END-DEFINE
...
SELECT * INTO PERS FROM SQL_PERSONNEL WHERE NAME = 'SCHMIDT'
COMPUTE AGE = AGE + 1
UPDATE PERS SET * WHERE CURRENT OF CURSOR
END-SELECT
...
```

**例 4 - assignment-list を使用した位置決め UPDATE**

```
DEFINE DATA LOCAL
1 PERS VIEW OF SQL-PERSONNEL
2 NAME
2 AGE
...
END-DEFINE
...
SELECT * INTO PERS FROM SQL-PERSONNEL WHERE NAME = 'SCHMIDT'
UPDATE SQL-PERSONNEL SET AGE = AGE + 1 WHERE CURRENT OF CURSOR
END-SELECT
...
```

# 151 参照プログラム例

---

▪ ASSIGN .....	1014
▪ AT BREAK .....	1015
▪ AT END OF DATA .....	1017
▪ AT END OF PAGE .....	1018
▪ AT START OF DATA .....	1018
▪ AT TOP OF PAGE .....	1020
▪ DEFINE SUBROUTINE .....	1021
▪ FIND .....	1022
▪ FOR .....	1024
▪ HISTOGRAM .....	1025
▪ IF .....	1025
▪ PERFORM BREAK PROCESSING .....	1027
▪ READ .....	1028
▪ REPEAT .....	1029
▪ SORT .....	1030
▪ STORE .....	1031
▪ UPDATE .....	1033
▪ システム変数のプログラム例 .....	1034

この章には、Naturalステートメントおよびシステム変数の参照ドキュメントで参照されている追加のプログラム例が含まれています。これらすべての例は SYSEXSYN ライブラリに含まれています。

## ASSIGN

---

次の例は、[ASSIGN/COMPUTE](#) ステートメントの説明で参照されています。

### ASGEX1R - ASSIGN (レポーティングモード)

```
** Example 'ASGEX1R': ASSIGN (reporting mode)
*****
RESET #A (N3)
      #B (A6)
      #C (N0.3)
      #D (N0.5)
      #E (N1.3)
      #F (N5)
      #G (A25)
      #H (A3/1:3)
*
#A = 5                                WRITE NOTITLE '=' #A
#B = 'ABC'                            WRITE '=' #B
#C = .45                              WRITE '=' #C
#D = #E = -0.12345                    WRITE '=' #D / '=' #E
ASSIGN ROUNDED #F = 199.999           WRITE '=' #F
#G = 'HELLO'                          WRITE '=' #G
*
#H (1) = 'UVW'
#H (3) = 'XYZ'                        WRITE '=' #H (1:3)
*
END
```

プログラム AEDEX1R の出力：

```
#A:    5
#B: ABC
#C:   .450
#D:  -.12345
#E: -0.123
#F:   200
#G: HELLO
#H: UVW      XYZ
```



## AT BREAK

次の例は、`AT BREAK` ステートメントの説明で参照されています。

### ATBEX1R - AT BREAK (レポートモード)

```
** Example 'ATBEX1R': AT BREAK (reporting mode)
*****
*
LIMIT 10
READ EMPLOYEES BY CITY
  AT BREAK OF CITY DO
    SKIP 1
  DOEND
/*
  DISPLAY NOTITLE CITY (IS=ON) COUNTRY (IS=ON) NAME
LOOP
END
```

プログラム ATBEX1R の出力：

CITY	COUNTRY	NAME
AIKEN	USA	SENKO
AIX EN OTHE	F	GODEFROY
AJACCIO		CANALE
ALBERTSLUND	DK	PLOUG
ALBUQUERQUE	USA	HAMMOND ROLLING FREEMAN LINCOLN
ALFRETON	UK	GOLDBERG
ALICANTE	E	GOMEZ

**ATBEX5R** - 複数のブレイクレベルがある **AT BREAK** ステートメント (レポートモード)

```

** Example 'ATBEX5R': AT BREAK (multiple break levels) (reporting mode)
*****
RESET LEAVE-DUE-L (N4)
*
LIMIT 5
FIND EMPLOYEES WITH CITY = 'PHILADELPHIA' OR = 'PITTSBURGH'
      SORTED BY CITY DEPT
MOVE LEAVE-DUE TO LEAVE-DUE-L
DISPLAY CITY (IS=ON) DEPT (IS=ON) NAME LEAVE-DUE-L
AT BREAK OF DEPT
  WRITE NOTITLE /
    T*DEPT OLD(DEPT) T*LEAVE-DUE-L SUM(LEAVE-DUE-L) /
AT BREAK OF CITY
  WRITE NOTITLE
    T*CITY OLD(CITY) T*LEAVE-DUE-L SUM(LEAVE-DUE-L) //
LOOP
*
END

```

プログラム ATBEX5R の出力：

CITY	DEPARTMENT CODE	NAME	LEAVE-DUE-L
PHILADELPHIA	MGMT30	WOLF-TERROINE	11
		MACKARNESS	27
	MGMT30		38
	TECH10	BUSH	39
		NETTLEFOLDS	24
TECH10		63	
PHILADELPHIA			101
PITTSBURGH	MGMT10	FLETCHER	34
	MGMT10		34
PITTSBURGH			34

## AT END OF DATA

次の例は、`AT END OF DATA` ステートメントの説明で参照されています。

### AEDEX1R - AT END OF DATA (レポートモード)

```
** Example 'AEDEX1R': AT END OF DATA (reporting mode)
*****
LIMIT 5
EMP. FIND EMPLOYEES WITH CITY = 'STUTTGART'
  IF NO RECORDS FOUND
    ENTER
  DISPLAY PERSONNEL-ID NAME FIRST-NAME
    SALARY (1) CURR-CODE (1)
/*
AT END OF DATA DO
  IF *COUNTER (EMP.) = 0 DO
    WRITE 'NO RECORDS FOUND'
    ESCAPE BOTTOM
  DOEND
  WRITE NOTITLE / 'SALARY STATISTICS:'
    / 7X 'MAXIMUM:' MAX(SALARY(1)) CURR-CODE (1)
    / 7X 'MINIMUM:' MIN(SALARY(1)) CURR-CODE (1)
    / 7X 'AVERAGE:' AVER(SALARY(1)) CURR-CODE (1)
  DOEND
LOOP
END
```

プログラム AEDEX1R の出力：

PERSONNEL ID	NAME	FIRST-NAME	ANNUAL SALARY	CURRENCY CODE
11100328	BERGHAUS	ROSE	70800	DM
11100329	BARTHEL	PETER	42000	DM
11300313	AECKERLE	SUSANNE	55200	DM
11300316	KANTE	GABRIELE	61200	DM
11500304	KLUGE	ELKE	49200	DM
SALARY STATISTICS:				
	MAXIMUM:	70800	DM	
	MINIMUM:	42000	DM	
	AVERAGE:	55680	DM	

## AT END OF PAGE

次の例は、`AT END OF PAGE` ステートメントの説明で参照されています。

### AEPEX1R - AT END OF PAGE (レポートモード)

```

** Example 'AEPEX1R': AT END OF PAGE (reporting mode)
*****
FORMAT PS=10
LIMIT 10
READ EMPLOYEES BY PERSONNEL-ID FROM '20017000'
  DISPLAY NOTITLE GIVE SYSTEM FUNCTIONS
    NAME JOB-TITLE 'SALARY' SALARY(1) CURR-CODE (1)
  /*
  AT END OF PAGE DO
    WRITE / 28T 'AVERAGE SALARY: ...' AVER(SALARY(1)) CURR-CODE (1)
  DOEND
  /*
LOOP
END

```

プログラム AEPEX1R の出力：

NAME	CURRENT POSITION	SALARY	CURRENCY CODE
CREMER	ANALYST	34000	USD
MARKUSH	TRAINEE	22000	USD
GEE	MANAGER	39500	USD
KUNEY	DBA	40200	USD
NEEDHAM	PROGRAMMER	32500	USD
JACKSON	PROGRAMMER	33000	USD
AVERAGE SALARY: ...		33533	USD

## AT START OF DATA

次の例は、`AT START OF DATA` ステートメントの説明で参照されています。

**ASDEX1R - AT START OF DATA** (レポートモード)

```
** Example 'ASDEX1R': AT START OF DATA (reporting mode)
*****
RESET #CITY (A20) #CNTL (A1)
*
REPEAT
  INPUT 'ENTER VALUE FOR CITY' #CITY
  /*
  IF #CITY = ' ' OR= 'END' DO
    STOP
  DOEND
  FIND EMPLOYEES WITH CITY = #CITY
  IF NO RECORDS FOUND DO
    WRITE NOTITLE NOHDR 'NO RECORDS FOUND'
    ESCAPE
  DOEND
  /*
  AT START OF DATA DO
    INPUT (AD=0) 'RECORDS FOUND' *NUMBER //
      'ENTER ''D'' TO DISPLAY RECORDS' #CNTL (AD=A)
    IF #CNTL NE 'D' DO
      ESCAPE BOTTOM
    DOEND
  DOEND
  /*
  DISPLAY NAME FIRST-NAME
  LOOP
LOOP
END
```

プログラム ASDEX1R の出力：

```
ENTER VALUE FOR CITY PARIS
```

市町村名を入力して確認した後：

```
RECORDS FOUND          26
ENTER 'D' TO DISPLAY RECORDS D
```

「D」を入力して確認した後：

NAME	FIRST-NAME
MAIZIERE	ELISABETH
MARX	JEAN-MARIE
REIGNARD	JACQUELINE
RENAUD	MICHEL
REMOUE	GERMAINE
LAVENDA	SALOMON
BROUSSE	GUY
GIORDA	LOUIS
SIECA	FRANCOIS
CENSIER	BERNARD
DUC	JEAN-PAUL
CAHN	RAYMOND
MAZUY	ROBERT
FAURIE	HENRI
VALLY	ALAIN
BRETON	JEAN-MARIE
GIGLEUX	JACQUES
KORAB-BRZOZOWSKI	BOGDAN
XOLIN	CHRISTIAN
LEGRIS	ROGER
VVVV	

## AT TOP OF PAGE

次の例は、`AT TOP OF PAGE` ステートメントの説明で参照されています。

### ATPEX1R - AT TOP OF PAGE (レポートニングモード)

```
** Example 'ATPEX1R': AT TOP OF PAGE (reporting mode)
*****
*
FORMAT PS=15
LIMIT 15
*
READ EMPLOYEES BY NAME STARTING FROM 'L'
  DISPLAY 2X NAME 4X FIRST-NAME CITY DEPT
  WRITE TITLE UNDERLINED 'EMPLOYEE REPORT'
  WRITE TRAILER '-' (78)
/*
  AT TOP OF PAGE DO
    WRITE 'BEGINNING NAME:' NAME
  DOEND
/*
  AT END OF PAGE DO
    SKIP 1
    WRITE 'ENDING NAME:  ' NAME
```

```
DOEND
LOOP
END
```

## DEFINE SUBROUTINE

次の例は、`DEFINE SUBROUTINE` ステートメントの説明で参照されています。

### DSREX1R - DEFINE SUBROUTINE (レポートモード)

```
** Example 'DSREX1R': DEFINE SUBROUTINE (reporting mode)
*****
RESET #ARRAY-ALL (A300)
      #X (N2) #Y (N2)
REDEFINE #ARRAY-ALL (#ARRAY (A75/1:4))
          #ARRAY-ALL (#ALINE (A25/1:4,1:3))
*
FORMAT PS=20
LIMIT 5
*
MOVE 1 TO #X #Y
*
FIND EMPLOYEES WITH NAME = 'SMITH'
  OBTAIN ADDRESS-LINE (1:2)
  /*
  MOVE NAME           TO #ALINE (#X,#Y)
  MOVE ADDRESS-LINE(1) TO #ALINE (#X+1,#Y)
  MOVE ADDRESS-LINE(2) TO #ALINE (#X+2,#Y)
  MOVE PHONE          TO #ALINE (#X+3,#Y)
  IF #Y = 3 DO
    MOVE 1 TO #Y
    PERFORM PRINT
  DOEND
  ELSE DO
    ADD 1 TO #Y
  DOEND
  AT END OF DATA DO
    PERFORM PRINT
  DOEND
LOOP
*
DEFINE SUBROUTINE PRINT
  WRITE NOTITLE (AD=OI) #ARRAY(*)
  RESET #ARRAY(*)
  SKIP 1
RETURN
*
END
```

プログラム AEDEX1R の出力：

```
SMITH          SMITH          SMITH
ENGLANDSVEJ 222 3152 SHETLAND ROAD 14100 ESWORTHY RD.
554349          MILWAUKEE          MONTERREY
                877-4563           994-2260

SMITH          SMITH
5 HAWTHORN     13002 NEW ARDEN COUR
OAK BROOK     SILVER SPRING
150-9351      639-8963
```

## FIND

---

次の例は、**FIND** ステートメントの説明で参照されています。

**FNDFIR - FIRST** オプションを指定した **FIND** ステートメント（レポートモード）

```
** Example 'FNDFIR': FIND FIRST
*****
*
FIND FIRST EMPLOYEES WITH CITY = 'DERBY'
*
WRITE NOTITLE 'TOTAL RECORDS SELECTED:' *NUMBER
SKIP 2
WRITE '***FIRST PERSON SELECTED***' //
      'NAME:      ' NAME /
      'DEPARTMENT:' DEPT /
      'JOB TITLE: ' JOB-TITLE
*
END
```

プログラム FNDFIR の出力：

```
TOTAL RECORDS SELECTED:      141

***FIRST PERSON SELECTED***

NAME:      DEAKIN
DEPARTMENT: SALE01
JOB TITLE: SALES ACCOUNTANT
```



**FNDNUM - NUMBER** オプションを指定した **FIND** ステートメント (レポートモード)

```

** Example 'FNDNUM': FIND NUMBER
*****
RESET #BIRTH (D)
*
MOVE EDITED '19500101' TO #BIRTH (EM=YYYYMMDD)
*
FIND NUMBER EMPLOYEES WITH CITY = 'MADRID'
      WHERE BIRTH LT #BIRTH
*
WRITE NOTITLE 'TOTAL RECORDS SELECTED:      ' *NUMBER
      / 'TOTAL BORN BEFORE 1 JAN 1950: ' *COUNTER
*
END

```

プログラム FNDNUM の出力：

```

TOTAL RECORDS SELECTED:      41
TOTAL BORN BEFORE 1 JAN 1950:  16

```

**FNDUNQ - UNIQUE** オプションを指定した **FIND** ステートメント (レポートモード)

```

** Example 'FNDUNQ': FIND UNIQUE
*****
RESET #NAME (A20)
*
*
INPUT 'ENTER EMPLOYEE NAME: ' #NAME
IF #NAME = ' '
  STOP
*
FIND UNIQUE EMPLOYEES WITH NAME = #NAME
*
DISPLAY NOTITLE NAME FIRST-NAME JOB-TITLE
*
ON ERROR DO
  WRITE 'NAME EITHER NOT UNIQUE OR DOES NOT EXIST'
  FETCH 'FNDUNQ'
DOEND
*
END

```

プログラム FNDUNQ の出力：

## 参照プログラム例

---

```
ENTER EMPLOYEE NAME: HEURTEBISE
```

「HEURTEBISE」という名前を入力して確認した後：

NAME	FIRST-NAME	CURRENT POSITION
HEURTEBISE	MICHEL	CONTROLEUR DE GESTION

## FOR

---

次の例は、[FOR](#) ステートメントの説明で参照されています。

### FOREX1R - FOR (レポートモード)

```
** Example 'FOREX1R': FOR (reporting mode)
*****
RESET #INDEX (I1)
      #ROOT (N2.7)
*
FOR #INDEX 1 TO 5
  COMPUTE #ROOT = SQRT (#INDEX)
  WRITE NOTITLE '=' #INDEX 3X '=' #ROOT
LOOP
*
SKIP 1
FOR #INDEX 1 TO 5 STEP 2
  COMPUTE #ROOT = SQRT (#INDEX)
  WRITE '=' #INDEX 3X '=' #ROOT
LOOP
*
END
```

プログラム FOREX1R の出力：

```
#INDEX:    1  #ROOT:    1.0000000
#INDEX:    2  #ROOT:    1.4142135
#INDEX:    3  #ROOT:    1.7320508
#INDEX:    4  #ROOT:    2.0000000
#INDEX:    5  #ROOT:    2.2360679

#INDEX:    1  #ROOT:    1.0000000
#INDEX:    3  #ROOT:    1.7320508
#INDEX:    5  #ROOT:    2.2360679
```

## HISTOGRAM

---

次の例は、**HISTOGRAM** ステートメントの説明で参照されています。

### HSTEX1R - HISTOGRAM (レポートモード)

```
** Example 'HSTEX1R': HISTOGRAM (reporting mode)
*****
*
LIMIT 8
HISTOGRAM EMPLOYEES CITY STARTING FROM 'M'
  DISPLAY NOTITLE CITY
    'NUMBER OF/PERSONS' *NUMBER *COUNTER
LOOP
*
END
```

プログラム HSTEX1R の出力：

CITY	NUMBER OF PERSONS	CNT
MADISON	3	1
MADRID	41	2
MAILLY LE CAMP	1	3
MAMERS	1	4
MANSFIELD	4	5
MARSEILLE	2	6
MATLOCK	1	7
MELBOURNE	2	8

## IF

---

次の例は、**IF** ステートメントの説明で参照されています。

IFEX1R - IF (レポーティングモード)

```

** Example 'IFEX1R': IF (reporting mode)
*****
RESET #BIRTH (D)
*
MOVE EDITED '19450101' TO #BIRTH (EM=YYYYMMDD)
SUSPEND IDENTICAL SUPPRESS
LIMIT 20
*
FND. FIND EMPLOYEES WITH CITY = 'FRANKFURT'
      SORTED BY NAME BIRTH
  IF SALARY (1) LT 40000
    WRITE NOTITLE '*****' NAME 30X 'SALARY LT 40000'
  ELSE DO
    IF BIRTH GT #BIRTH DO
      FIND VEHICLES WITH PERSONNEL-ID = PERSONNEL-ID (FND.)
      DISPLAY (IS=ON) NAME BIRTH (EM=YYYY-MM-DD)
      SALARY (1) MAKE (AL=8)
    LOOP
  DOEND
DOEND
LOOP
END

```

プログラム IFEX1R の出力：

NAME	DATE OF BIRTH	ANNUAL SALARY	MAKE
BAECKER	1956-01-05	74400	BMW
***** BECKER			SALARY LT 40000
BLOEMER	1979-11-07	45200	FIAT
FALTER	1954-05-23	70800	FORD
***** FALTER			SALARY LT 40000
***** GROTHE			SALARY LT 40000
***** HEILBROCK			SALARY LT 40000
***** HESCHMANN			SALARY LT 40000
HUCH	1952-09-12	67200	MERCEDES
***** KICKSTEIN			SALARY LT 40000
***** KLEENE			SALARY LT 40000
***** KRAMER			SALARY LT 40000

## PERFORM BREAK PROCESSING

次の例は、`PERFORM BREAK PROCESSING` ステートメントの説明で参照されています。

### PBPEX1R - PERFORM BREAK PROCESSING (レポートモード)

```
** Example 'PBPEX1R': PERFORM BREAK PROCESSING (reporting mode)
*****
RESET #LINE (N2) #INDEX (N2)
*
MOVE 1 TO #LINE
FOR #INDEX 1 TO 18
  PERFORM BREAK PROCESSING
  /*
  AT BREAK OF #INDEX /1/ DO
    WRITE NOTITLE / 'PLEASE COMPLETE LINES 1-9 ABOVE' /
    MOVE 1 TO #LINE
  DOEND
  /*
  WRITE NOTITLE '_' (64) '=' #LINE
  ADD 1 TO #LINE
LOOP
END
```

プログラム PBPEX1R の出力：

```
_____ #LINE: 1
_____ #LINE: 2
_____ #LINE: 3
_____ #LINE: 4
_____ #LINE: 5
_____ #LINE: 6
_____ #LINE: 7
_____ #LINE: 8
_____ #LINE: 9

PLEASE COMPLETE LINES 1-9 ABOVE

_____ #LINE: 1
_____ #LINE: 2
_____ #LINE: 3
_____ #LINE: 4
_____ #LINE: 5
_____ #LINE: 6
_____ #LINE: 7
_____ #LINE: 8
_____ #LINE: 9

PLEASE COMPLETE LINES 1-9 ABOVE
```

## READ

次の例は、[READ](#) ステートメントの説明で参照されています。

### REAEX1R - READ (レポートニングモード)

```

** Example 'REAEX1R': READ (reporting mode)
*****
LIMIT 3
*
WRITE 'READ IN PHYSICAL SEQUENCE'
READ EMPLOYEES IN PHYSICAL SEQUENCE
  DISPLAY NOTITLE PERSONNEL-ID NAME *ISN *COUNTER
LOOP
*
WRITE / 'READ IN ISN SEQUENCE'
READ EMPLOYEES BY ISN STARTING FROM 1 ENDING AT 3
  DISPLAY PERSONNEL-ID NAME *ISN *COUNTER
LOOP
*
WRITE / 'READ IN NAME SEQUENCE'
READ EMPLOYEES BY NAME
  DISPLAY PERSONNEL-ID NAME *ISN *COUNTER
LOOP
*
WRITE / 'READ IN NAME SEQUENCE STARTING FROM 'M''
READ EMPLOYEES BY NAME STARTING FROM 'M'
  DISPLAY PERSONNEL-ID NAME *ISN *COUNTER
LOOP
*
END

```

プログラム REAEX1R の出力：

PERSONNEL ID	NAME	ISN	CNT
-----			
READ IN PHYSICAL SEQUENCE			
50005800	ADAM	1	1
50005600	MORENO	2	2
50005500	BLOND	3	3
READ IN ISN SEQUENCE			
50005800	ADAM	1	1
50005600	MORENO	2	2
50005500	BLOND	3	3
READ IN NAME SEQUENCE			

```

60008339 ABELLAN 478 1
30000231 ACHIESON 878 2
50005800 ADAM 1 3

READ IN NAME SEQUENCE STARTING FROM 'M'
30008125 MACDONALD 923 1
20028700 MACKARNESS 765 2
40000045 MADSEN 508 3

```

## REPEAT

次の例は、**REPEAT** ステートメントの説明で参照されています。

### RPTEX1R - REPEAT (レポートモード)

```

** Example 'RPTEX1R': REPEAT (reporting mode)
*****
RESET #PERS-NR (A8)
*
REPEAT
  INPUT 'ENTER A PERSONNEL NUMBER:' #PERS-NR
  IF #PERS-NR = ' '
    ESCAPE BOTTOM
  FIND EMPLOYEES WITH PERSONNEL-ID = #PERS-NR
  IF NO RECORD FOUND
    REINPUT 'NO RECORD FOUND'
  DISPLAY NOTITLE NAME
  LOOP
LOOP
*
END

```

プログラム RPTEX1R の出力：

```

ENTER A PERSONNEL NUMBER:

```

### RPTEX2R - WHILE および UNTIL オプションを指定した REPEAT (レポートモード)

```

** Example 'RPTEX2R': REPEAT (with WHILE and UNTIL option)
*****
RESET #X (I1) #Y (I1)
*
*
REPEAT WHILE #X <= 5
  ADD 1 TO #X
  WRITE NOTITLE '=' #X
  LOOP
*

```

```
SKIP 3
REPEAT
  ADD 1 TO #Y
  WRITE '=' #Y
  UNTIL #Y = 6
LOOP
*
END
```

プログラム RPTEX2R の出力：

```
#X: 1
#X: 2
#X: 3
#X: 4
#X: 5
#X: 6
```

```
#Y: 1
#Y: 2
#Y: 3
#Y: 4
#Y: 5
#Y: 6
```

## SORT

---

次の例は、[SORT](#) ステートメントの説明で参照されています。

### SRTEX1R - SORT (レポートモード)

```
** Example 'SRTEX1R': SORT (reporting mode)
*****
RESET #AVG (P11) #TOTAL-TOTAL (P11) #TOTAL-SALARY (P11)
      #AVER-PERCENT (N3.2)
*
LIMIT 3
FIND EMPLOYEES WITH CITY = 'BOSTON'
  OBTAIN SALARY(1:2)
  COMPUTE #TOTAL-SALARY = SALARY (1) + SALARY (2)
  ACCEPT IF #TOTAL-SALARY GT 0
  /*
  SORT BY PERSONNEL-ID USING #TOTAL-SALARY SALARY(*) CURR-CODE
      GIVE AVER(#TOTAL-SALARY)
  /*
  AT START OF DATA DO
```



```

WRITE NOTITLE '*' (40)
      'AVG CUMULATIVE SALARY:' *AVER (#TOTAL-SALARY) /
MOVE *AVER (#TOTAL-SALARY) TO #AVG
DOEND
COMPUTE ROUNDED #AVER-PERCENT = #TOTAL-SALARY / #AVG * 100
ADD #TOTAL-SALARY TO #TOTAL-TOTAL
/*
DISPLAY NOTITLE PERSONNEL-ID SALARY (1) SALARY (2)
      #TOTAL-SALARY CURR-CODE (1)
      'PERCENT/OF/AVER' #AVER-PERCENT
AT END OF DATA
      WRITE / '*' (40) 'TOTAL SALARIES PAID: ' #TOTAL-TOTAL
LOOP
*
END

```

プログラム SRTEX1R の出力：

PERSONNEL ID	ANNUAL SALARY	ANNUAL SALARY	#TOTAL-SALARY	CURRENCY CODE	PERCENT OF AVER
***** AVG CUMULATIVE SALARY:					44633
20000100	31000	29400	60400	USD	135.30
20019200	18000	17100	35100	USD	78.60
20020400	20000	18400	38400	USD	86.00
***** TOTAL SALARIES PAID:					133900

## STORE

次の例は、[STORE](#) ステートメントの説明で参照されています。

### STOEX1R - STORE (レポートモード)

```

** Example 'STOEX1R': STORE (reporting mode)
**
** CAUTION: Executing this example will modify the database records!
*****
RESET #PERSONNEL-ID (A8)
      #NAME           (A20)
      #FIRST-NAME    (A15)
      #BIRTH-D       (D)
      #MAR-STAT      (A1)
      #BIRTH         (A8)
      #CITY          (A20)

```

```

        #COUNTRY      (A3)
        #CONF         (A1)
*
REPEAT
  INPUT 'ENTER A PERSONNEL ID AND NAME (OR ''END'' TO END)' //
        'PERSONNEL-ID : ' #PERSONNEL-ID //
        'NAME          : ' #NAME          /
        'FIRST-NAME    : ' #FIRST-NAME
  /*
  /*  VALIDATE ENTERED DATA
  /*
  IF #PERSONNEL-ID = 'END' OR #NAME = 'END'
    STOP
  IF #NAME = ' '
    REINPUT WITH TEXT 'ENTER A LAST-NAME' MARK 2 AND SOUND ALARM
  IF #FIRST-NAME = ' '
    REINPUT WITH TEXT 'ENTER A FIRST-NAME' MARK 3 AND SOUND ALARM
  /*
  /*  ENSURE PERSON IS NOT ALREADY ON FILE
  /*
  FIND NUMBER EMPLOYEES WITH PERSONNEL-ID = #PERSONNEL-ID
  IF *NUMBER > 0
    REINPUT 'PERSON WITH SAME PERSONNEL-ID ALREADY EXISTS'
          MARK 1 AND SOUND ALARM
  MOVE 'N' TO #CONF
  /*
  /*  GET FURTHER INFORMATION
  /*
  INPUT
    'ADDITIONAL PERSONNEL DATA'          ////
    'PERSONNEL-ID          : ' #PERSONNEL-ID (AD=IO) /
    'NAME                  : ' #NAME          (AD=IO) /
    'FIRST-NAME           : ' #FIRST-NAME   (AD=IO) ///
    'MARITAL STATUS       : ' #MAR-STAT     /
    'DATE OF BIRTH (YYYYMMDD) : ' #BIRTH      /
    'CITY                 : ' #CITY         /
    'COUNTRY (3 CHARACTERS) : ' #COUNTRY     //
    'ADD THIS RECORD (Y/N) : ' #CONF        (AD=M)
  /*
  /*  ENSURE REQUIRED FIELDS CONTAIN VALID DATA
  /*
  IF NOT (#MAR-STAT = 'S' OR = 'M' OR = 'D' OR = 'W')
    REINPUT TEXT 'ENTER VALID MARITAL STATUS S=SINGLE ' -
          'M=MARRIED D=DIVORCED W=WIDOWED' MARK 1
  IF NOT (#BIRTH = MASK(YYYYMMDD) AND #BIRTH = MASK(1582-2699))
    REINPUT TEXT 'ENTER CORRECT DATE' MARK 2
  IF #CITY = ' '
    REINPUT TEXT 'ENTER A CITY NAME' MARK 3
  IF #COUNTRY = ' '
    REINPUT TEXT 'ENTER A COUNTRY CODE' MARK 4
  IF NOT (#CONF = 'N' OR = 'Y')
    REINPUT TEXT 'ENTER Y (YES) OR N (NO)' MARK 5

```

```

IF #CONF = 'N'
  ESCAPE TOP
/*
/*  ADD THE RECORD
/*
MOVE EDITED #BIRTH TO #BIRTH-D (EM=YYYYMMDD)
/*
STORE RECORD IN EMPLOYEES
  WITH PERSONNEL-ID = #PERSONNEL-ID
      NAME           = #NAME
      FIRST-NAME     = #FIRST-NAME
      MAR-STAT       = #MAR-STAT
      BIRTH          = #BIRTH-D
      CITY           = #CITY
      COUNTRY        = #COUNTRY
END OF TRANSACTION
/*
WRITE NOTITLE 'RECORD HAS BEEN ADDED'
/*
LOOP
END

```

## UPDATE

次の例は、[UPDATE](#) ステートメントの説明で参照されています。

### UPDEX1R - UPDATE (レポーティングモード)

```

** Example 'UPDEX1R': UPDATE (reporting mode)
**
** CAUTION: Executing this example will modify the database records!
*****
RESET #NAME (A20)
*
INPUT 'ENTER A NAME:' #NAME (AD=M)
IF #NAME = ' '
  STOP
*
FIND EMPLOYEES WITH NAME = #NAME
IF NO RECORDS FOUND
  REINPUT WITH 'NO RECORDS FOUND' MARK 1
/*
INPUT 'NAME:          ' NAME (AD=0) /
      'FIRST NAME:' FIRST-NAME (AD=M) /
      'CITY:          ' CITY (AD=M)
/*
UPDATE USING SAME RECORD
/*
END TRANSACTION

```

## 参照プログラム例

---

```
/*  
LOOP  
*  
END
```

プログラム UPDEX1R の出力：

```
ENTER A NAME:
```

## システム変数のプログラム例

---

次の例は、\*OCCURRENCE システム変数の説明で参照されています。

### OCC1P - システム変数 \*OCCURRENCE

```
** Example 'OCC1P': *OCCURRENCE  
*****  
DEFINE DATA LOCAL  
1 #N1 (N7/1:10)  
1 #N2 (N7/1:10,1:10)  
1 #N3 (N7/1:10,1:10,1:10)  
END-DEFINE  
*  
CALLNAT 'OCC1N' #N1(*) #N2(1:2,1:4) #N3(1:6,1:7,1:8)  
*  
END
```

プログラム OCC1P で呼び出されるサブプログラム OCC1N：

```
** Example 'OCC1N': *OCCURRENCE (called by OCC1P)  
*****  
DEFINE DATA  
PARAMETER  
1 PARM1 (N7/1:V)  
1 PARM2 (N7/1:V,1:V)  
1 PARM3 (N7/1:V,1:V,1:V)  
LOCAL  
1 #OCC2 (I4/1:2)  
1 #OCC3 (I4/1:3)  
1 #OCC1 (I4)  
END-DEFINE  
*  
MOVE *OCC(PARM1) TO #OCC1  
MOVE *OCC(PARM2,*) TO #OCC2(*)  
MOVE *OCC(PARM3,*) TO #OCC3(*)  
*  
DISPLAY #OCC1 #OCC2(*) #OCC3(*)  
DISPLAY *OCC(PARM1,*) *OCC(PARM2,*) *OCC(PARM3,*)
```

```

*
NEWPAGE
*
WRITE NOHDR
  'Occurrences of 1. parameter:' *OCC(PARM1)
  / 'Occurrences of 1. parameter:' *OCC(PARM1,1)
  / 'Occurrences of 1. parameter:' *OCC(PARM1,*)
  / 'Occurrences of 2. parameter:' *OCC(PARM2,1) *OCC(PARM2,2)
  / 'Occurrences of 2. parameter:' *OCC(PARM2,*)
  / 'Occurrences of 3. parameter:' *OCC(PARM3,1) *OCC(PARM3,2)
  / 'Occurrences of 3. parameter:' *OCC(PARM3,3)
  / 'Occurrences of 3. parameter:' *OCC(PARM3,*)
*
END

```

プログラム OCC1P の出力 - ページ 1 :

```

Page      1                                05-01-18  10:21:30
-----
#OCC1      #OCC2      #OCC3
-----
          10           2           6
                   4           7
                   8
          10           2           6
                   4           7
                   8

```

プログラム OCC1P の出力 - ページ 2 :

```

Page      2                                05-01-18  10:21:30
Occurrences of 1. parameter:      10
Occurrences of 1. parameter:      10
Occurrences of 1. parameter:      10
Occurrences of 2. parameter:       2           4
Occurrences of 2. parameter:       2           4
Occurrences of 3. parameter:       6           7           8
Occurrences of 3. parameter:       6           7           8

```

**OCC2P - システム変数 \*OCCURRENCE**

```
** Example 'OCC2P': *OCCURRENCE
*****
DEFINE DATA LOCAL
1 #N (N7/1:10)
1 #I (I4)
END-DEFINE
*
FOR #I=1 TO 10
  MOVE #I TO #N(#I)
END-FOR
*
WRITE 'Passing ocurrences 1:5'
CALLNAT 'OCC2N' #N(1:5)
*
WRITE 'Passing ocurrences 5:10'
CALLNAT 'OCC2N' #N(5:10)
*
END
```

プログラム OCC2P で呼び出されるサブプログラム OCC2N :

```
** Example 'OCC2N': *OCCURRENCE (called by OCC2P)
*****
DEFINE DATA
PARAMETER
1 #ARR (N7/1:V)
LOCAL
1 I (N7)
END-DEFINE
*
FOR I=1 TO *OCC(#ARR)
  DISPLAY #ARR(I)
END-FOR
*
END
```

プログラム OCC2P の出力 :

```
Page      1                                05-01-18  10:33:03

Passing ocurrences 1:5
  1
  2
  3
  4
  5
Passing ocurrences 5:10
  5
```

6  
7  
8  
9  
10





# 索引

---

## す

ステートメント, 1

